




Д. П. Голоскоков

ПРАКТИЧЕСКИЙ КУРС МАТЕМАТИЧЕСКОЙ ФИЗИКИ В СИСТЕМЕ MAPLE



УЧЕБНОЕ ПОСОБИЕ
ДЛЯ ВУЗОВ

- Для студентов и преподавателей высших учебных заведений
- Обыкновенные дифференциальные уравнения
- Уравнения в частных производных
- Метод Фурье
- Интегральные преобразования
- Численные и приближенные аналитические методы решения задач математической физики
- Математическая теория поля
- Примеры решения типовых задач на компьютере и задачи для самостоятельного решения

Предисловие

Д. П. Голоскоков

ПРАКТИЧЕСКИЙ КУРС МАТЕМАТИЧЕСКОЙ ФИЗИКИ В СИСТЕМЕ MAPLE

*Рекомендовано Учебно-методическим
объединением высших учебных заведений
Российской Федерации по образованию
в области гидрометеорологии
в качестве учебного пособия
для студентов высших учебных заведений*

Санкт-Петербург
2010

УДК 517.958(075)

ББК 22.162.6я7

Г61

Рецензенты:

Доктор физико-математических наук, профессор *Ю. М. Даль* (СПбГУ)
Доктор физико-математических наук, профессор, *М. Н. Курсанов* (МЭИ (ТУ))

Голоскоков Д. П.

Г61 Практический курс математической физики в системе Maple:
учеб. пособие для вузов. — СПб.: ООО «ПаркКом», 2010. — 643 с.

ISBN 978-5-98175-034-2

Книга является оригинальным и полезным дополнением к учебным курсам по системе компьютерной математики Maple для Windows. Отличительная ее особенность состоит в том, что она посвящена основным методам решения задач математической физики. Чтобы сделать книгу доступной широкому кругу читателей, все необходимые теоретические сведения представлены наиболее просто и наглядно, избегая излишних обобщений и абстрактных понятий.

Основное внимание уделяется развитию навыков конкретного применения тех или иных аналитических или численных методов решения задач математической физики. В конце каждой главы приведены задачи. Все они снабжены ответами, а некоторые из них — указаниями. В последней главе достаточно подробно рассмотрены задачи теории поля, интересные не только в курсе математической физики для студентов III и IV курсов, но и младших курсов, изучающих математический анализ.

Учебное пособие может быть также полезно аспирантам, инженерам и научным сотрудникам, специализирующимся в области прикладной математики.

УДК 517.958(075)

ББК 22.162.6я7

ISBN 978-5-98175-034-2

© Д. П. Голоскоков, 2010
© ООО «ПаркКом», 2010

Предисловие

Настоящий учебный курс представляет собой учебное пособие – практикум по методам математической физики и имеет целью дать необходимый материал для приобретения практических навыков в применении этих методов и системы Maple для решения различных задач математической физики. Книга включает в себя краткие теоретические сведения, задачи и упражнения по следующим специальным разделам курса высшей математики: «Обыкновенные дифференциальные уравнения», «Классификация уравнений математической физики», «Метод Фурье», «Метод собственных функций в применении к неоднородным задачам математической физики», «Задачи математической физики с непрерывным спектром. Интегральные преобразования», «Численные методы решения задач математической физики», «Методы взвешенных невязок» и «Математическая теория поля».

В учебном пособии рассматриваются типовые задачи с подробными решениями и пояснениями. Приводится достаточное количество задач для самостоятельного решения с ответами. К отдельным задачам даются методические указания. Данное пособие является дополнением к учебнику автора: Уравнения математической физики. Решение задач в системе Maple. Учебник для вузов. — СПб.: Питер, 2004, 539 с.

Особенностью книги является широкое использование системы аналитических вычислений Maple при решении учебных задач математической физики. В первой, вводной главе книги кратко рассмотрены основные элементы системы Maple, знакомство с которыми необходимо при ее использовании для решения задач. Более подробно используемые команды Maple рассматриваются непосредственно при решении конкретных задач математической физики; описаны новые возможности системы Maple применительно к рассматриваемым задачам.

С появлением и доступностью персональных компьютеров, а самое главное, с появлением мощных систем аналитических вычислений, таких как Maple, пожалуй, в наибольшей степени изменился характер, и повысилась производительность умственного труда. В настоящее время все инженерные, конструкторские, экономические задачи можно решать на компьютере, причем в подавляющем большинстве случаев совершенно нет необходимости заниматься программированием в традиционном смысле. Например, в системе аналитических вычислений Maple пользователь имеет возможность выполнить все расчеты так, как он выполнил бы их на листе бумаги, причем все рутинные и трудоемкие вычисления (а главное, без ошибок и в формульном, аналитическом виде!) берет на себя система Maple.

Пакет Maple ныне широко используется для преподавания математики и технических дисциплин во многих учебных заведениях мира, в том числе и России. Для студентов Maple является неоценимым помощником в изучении разнообразных математических методов, освобождая их от рутинных математических вычислений и сосредотачивая внимание на существе изучаемого метода. В этой книге приводятся решения типовых задач с использованием замечательных возможностей пакета Maple.

Пособие не ставит целью изучение самой системы аналитических вычислений Maple. С одной стороны, в настоящее время имеется уже достаточное количество различных книг по системе Maple. С другой стороны, при использовании большинства таких книг, не имеющих проблемной ориентации, у пользователей (и не только студентов!) со своей, специфической, сферой профессиональных интересов при изучении системы возникают трудности. В отличие от таких книг круг рассматриваемых в настоящей книге приемов использования, функций и команд Maple не является всеобъемлющим, а ориентирован, прежде всего, на решение задач математической физики. Все средства, понятия, функции и команды Maple рассматриваются по мере необходимости их использования в связи с конкретной задачей. Таким образом, читателю, которому необходимо быстро и без ошибок выполнить сложное задание по математической физике, нет необходимости предварительно знакомиться с принципами работы в Maple — знакомство с системой происходит в процессе решения той или иной задачи.

Система Maple бесспорно занимает одно из лидирующих мест среди систем компьютерной алгебры, или, как говорят, символьных или аналитических вычислений, и обеспечивает пользователю удобную и интеллектуальную среду для математического исследования.

Система Maple состоит из ядра — процедур, написанных на языке «С» и в высшей степени оптимизированных, библиотеки, написанной на Maple-языке и интерфейса. Ядро выполняет большинство базисных операций. Библиотека содержит большое количество команд-процедур, выполняемых в режиме интерпретации. Программируя собственные процедуры, «продвинутый» пользователь может пополнять ими стандартный набор и, таким образом, значительно расширять возможности Maple. Следует отметить, что возможности Maple настолько разнообразны, что подавляющее большинство задач решаются стандартными средствами Maple, и для большинства пользователей нет необходимости в написании своих собственных программ.

Система Maple достаточно легко осваивается, удобна в работе, так что ее может использовать даже школьник для простых расчетов или для освоения математики. В то же время система обладает настолько обширным набором функций и вычислительных средств, что она с успехом может быть применена для профессиональной работы в области математики и смежных дисциплин. При постепенном освоении системы Maple возникает ощущение, что ваши способности к точным наукам возрастают!

Книга рассчитана на студентов университетов и других учебных заведений, программы которых включают соответствующие главы курса математической физики, а также на аспирантов и научных работников, сталкивающихся в своей работе с необходимостью выполнения расчетов, требующих применения методов математической физики.

Список литературы, главным образом, задачников, которые наиболее широко использовались, дан в конце книги.

Д. Голоскоков

1. ВВЕДЕНИЕ В MAPLE

Графический интерфейс пользователя

- Общий вид
- Рабочие листы
- Область ввода
- Область вывода
- Вывод графики

Работа с меню

- Виды меню
- Основное меню
 - Команды меню File*
 - Команды меню Edit*
 - Команды меню View*
 - Команды меню Window*

Справочная система

- Организация справки
- Вызов справки с помощью меню Help
- Вызов справки из рабочего листа

Основы работы

- Объекты, переменные и выражения
- Команды
- Команды и функции ядра
 - Команда simplify()*
 - Команда expand()*
 - Команда factor()*
 - Команда combine()*
 - Команда collect()*
 - Команда normal()*
 - Команда assume()*
 - Команда subs()*
- Сложные типы данных
 - Последовательность, список и множество*
 - Массивы и таблицы*
 - Вычисление выражений и переменных сложных типов*
- Решение уравнений, неравенств и их систем
 - Команда solve()*
 - Команда fsolve()*
- Дифференцирование и интегрирование

Пакеты

- Графика
 - Команды двумерной графики*

Элементы программирования в Maple

- Функции и процедуры
- Условный оператор
- Оператор цикла

2. ОБЫКНОВЕННЫЕ ДИФФЕРЕНЦИАЛЬНЫЕ УРАВНЕНИЯ

Общие интегралы дифференциальных уравнений и систем

Задача Коши и краевая задача для обыкновенных дифференциальных уравнений

- Функция Грина

Линейные дифференциальные уравнения с постоянными коэффициентами
Задача Штурма-Лиувилля

Упражнения

3. КЛАССИФИКАЦИЯ УРАВНЕНИЙ В ЧАСТНЫХ ПРОИЗВОДНЫХ ВТОРОГО ПОРЯДКА

Классификация уравнений в точке

Классификация уравнений с двумя независимыми переменными

Преобразование уравнений второго порядка с помощью замены переменных

Уравнение гиперболического типа

Уравнение параболического типа

Уравнение эллиптического типа

Понятие об общем интеграле уравнения в частных производных

Примеры

Применение общего интеграла к решению некоторых задач математической физики

Колебания неограниченной струны. Метод Даламбера

Обобщение формулы Даламбера на многомерные задачи

Примеры решения типовых задач

Упражнения

4. МЕТОД ФУРЬЕ

Уравнения с разделяющимися переменными

Метод Фурье для случая двух независимых переменных

Примеры решения типовых задач

Упражнения

5. НЕОДНОРОДНЫЕ ЗАДАЧИ МАТЕМАТИЧЕСКОЙ ФИЗИКИ

Метод приведения к однородной задаче

Метод Гринберга

Краткая схема решения задачи методом Гринберга

Примеры решения типовых задач

Упражнения

6. ЗАДАЧИ МАТЕМАТИЧЕСКОЙ ФИЗИКИ С НЕПРЕРЫВНЫМ СПЕКТРОМ. ИНТЕГРАЛЬНЫЕ ПРЕОБРАЗОВАНИЯ

Некоторые интегральные разложения, связанные с сингулярной задачей

Штурма-Лиувилля

Метод интегральных преобразований

Преобразование Фурье

Преобразование Лапласа

Формула обращения

Неоднородные задачи с непрерывным спектром

Примеры решения типовых задач

Упражнения

7. ЧИСЛЕННЫЕ МЕТОДЫ РЕШЕНИЯ ЗАДАЧ МАТЕМАТИЧЕСКОЙ ФИЗИКИ

Численное решение обыкновенных дифференциальных уравнений

Задача Коши

Метод Эйлера

Метод Рунге-Кутты

Краевая задача

Метод пристрелки

Метод конечных разностей

Численное решение дифференциальных уравнений в частных производных
Упражнения

8. МЕТОДЫ ВЗВЕШЕННЫХ НЕВЯЗОК

Аппроксимация функций

Поточечная коллокация
Коллокация по подобластям
Метод Галеркина

Аппроксимация решений дифференциальных уравнений

Аппроксимация решений дифференциальных уравнений и краевых условий

Естественные краевые условия

Естественные краевые условия для двумерного стационарного уравнения теплопроводности

Методы граничного решения

Нелинейные задачи

Упражнения

9. МАТЕМАТИЧЕСКАЯ ТЕОРИЯ ПОЛЯ

Скалярное поле

Примеры

Производная скалярного поля по направлению

Примеры

Градиент скалярного поля

Примеры

Векторное поле

Примеры

Поток векторного поля. Формула Остроградского

Примеры

Дивергенция векторного поля

Примеры

Ротор векторного поля

Примеры

Линейный интеграл и циркуляция векторного поля. Формула Стокса

Примеры

Потенциальное и соленоидальное векторные поля

Примеры

Дифференциальные операции второго порядка

Примеры

Упражнения

Введение в Maple

Пакет Maple является системой аналитических вычислений, предназначенной для решения математических задач, а, значит, инженерных, финансовых, экономических, технических и других задач, требующих привлечения разнообразного математического аппарата. Пользователем данной системы может быть любой человек, владеющий математикой в пределах необходимых ему знаний для решения задач проблемной области. Это может быть инженер, занимающийся расчетом конструкций, финансист, следящий за потоками денежных масс на предприятии, математик, использующий Maple для облегчения своей работы, и т.д.

Пакет Maple работает под управлением операционной системы Windows, и его интерфейс пользователя является стандартным для программ, разработанных для выполнения под управлением этой операционной системы: меню, панели инструментов, рабочая область, возможность загрузки нескольких документов Maple в одном сеансе работы и т.п.

Пользователь с помощью клавиатуры набирает команды (функции) в рабочем поле, которые передаются на обработку основному компоненту системы, называемому ядром системы. Названия, или имена команд и функций соответствуют тем действиям, которые они выполняют, что достаточно удобно. Любая, введенная пользователем правильная команда Maple, немедленно интерпретируется исполняющей системой, и пользователь сразу же видит результат.

Кроме непосредственного ввода последовательности необходимых команд для решения задачи Maple предоставляет собственный язык программирования, операторы которого похожи на операторы любого языка программирования высокого уровня. Это позволяет создавать собственную последовательность действий для решения конкретной часто выполняемой задачи, и оформить ее в виде процедуры, которую впоследствии можно вызывать в любое время.

Изучив небольшой набор команд и правил, можно приступить к работе в системе Maple. Справочная система поможет в дальнейшем освоении и изучении данной системы аналитических вычислений.

Графический интерфейс пользователя

Любое Windows-приложение взаимодействует с пользователем через свой интерфейс, который представляет собой «окно» с расположенными в нем разнообразными стандартными компонентами: меню, панелями инструментов, полями ввода и т.д. Этот параграф посвящен описанию основных компонентов интерфейса пользователя системы Maple, которому мы уделим столько времени, сколько необходимо для того, чтобы начать работу. Наша основная цель — научить решать задачи с помощью Maple, а умение работать с интерфейсом придет по мере накопления опыта работы с системой.

Общий вид

Основной компонент Maple — рабочий лист (Worksheet), в который пользователь заносит не только команды Maple, но и комментарии в виде форматированного

текста. Развитая система графики позволяет встраивать в рабочий лист двумерные и трехмерные графические отображения, поясняющие и помогающие решению конкретной поставленной задачи.

Начиная с версии Maple 9 и старше, после установки пакета на компьютер на рабочем столе появляются два ярлыка — Maple 9 (соответственно Maple 10, Maple 11, Maple 12 и т.д.) и Classic Worksheet Maple 9 (соответственно 10, 11 12 и т.д.), соответствующие двум вариантам загрузки программы. Далее мы не будем выделять какую-либо версию Maple. Наша цель научиться использовать Maple на практике, а какая это версия — не важно! Понятно, что с каждой новой версией возможности системы возрастают. С этими возможностями можно познакомиться по справочной системе Maple.

Итак, пакет Maple предоставляет пользователю два варианта интерфейса рабочего листа, которые оба имеют полный доступ ко всему математическому аппарату системы и предоставляют все возможности новой версии (начиная с 9 и старше). По умолчанию загружается усовершенствованный вариант интерфейса рабочего листа Maple. Классический вариант рабочего листа Classic Worksheet Maple, доступный для Windows и Unix платформ и привычный для пользователя по ранним версиям Maple, имеет традиционный для ранних версий вид рабочего листа; требует при загрузке меньше памяти и рекомендуется для компьютера с небольшим объемом оперативной памяти. В Windows пользователь может изменить ассоциацию Maple-файла, закрепив его за тем или иным вариантом интерфейса рабочего листа. Это можно сделать, запустив Worksheet File Association Selector (Пуск → Все программы → Maple → Tools → Worksheet File Association Selector). Как утверждают разработчики системы Maple, новый стандарт интерфейса рабочего листа предоставляет большие возможности форматирования с использованием различных шрифтов и цветовой гаммы, лучшую организацию меню, более простое управление секциями и подсекциями и другие улучшения.

Система Maple имеет стандартный оконный интерфейс пользователя для программ, работающих под управлением Windows. При загрузке программы на экране монитора отображается окно интерфейса. В нем можно выделить несколько основных компонентов (частей).

В верхней части, непосредственно под заголовком окна, расположена *строка меню*, в которой находятся меню системы Maple. При нажатии левой кнопкой мыши на каком-нибудь из меню, например File (Файл), раскрывается окно, в котором находятся команды и подменю данного меню.

Ниже полосы меню расположена *основная панель инструментов* с рядом кнопок, дублирующих наиболее часто используемые команды меню. Щелчок мыши на кнопке приводит к выполнению ассоциированной с ней команды.

Большую часть окна интерфейса занимает *рабочая область*, в которой вводятся команды и отображаются результаты их выполнения.

Непосредственно под основной панелью инструментов расположена *контекстная панель инструментов*, вид которой зависит от того, в какой области рабочего листа расположен курсор. Существует пять видов контекстных панелей инструментов: для выделенного двумерного графика, для выделенного трехмерного графика, для

выделенной анимационной графики, для области вывода и для области ввода рабочего листа.

В нижней части интерфейса расположена *строка состояния*, в которой отображаются некоторые параметры, относящиеся к исполняющей системе Maple, а также краткая информация относительно выбранной команды меню.

Во время работы для определенных объектов рабочего листа можно отобразить набор команд, применимых к данным объектам. Для этого достаточно расположить курсор над объектом и нажать правую кнопку мыши. Появится контекстно-зависимое меню для этого объекта (небольшая панель с набором команд).

Рабочие листы

Технология работы в Maple представляет собой интерактивный сеанс: пользователь вводит команды в рабочем поле интерфейса и передает их на выполнение исполняющей системе (ядру) Maple. Все вводимые команды и отображаемые результаты вычислений представляют собой *рабочий лист* — основной документ, который создает Maple и с которым он работает. При завершении сеанса работы его можно сохранить на диске в файлах разных форматов, а при очередном сеансе открыть и снова выполнить все команды, содержащиеся в нем, или произвести его корректировку.

Рабочий лист состоит из *области ввода* и *области вывода*. В первой пользователь вводит команды Maple, которые передаются на выполнение в ядро системы. В области вывода отображаются результаты выполнения команд и операторов языка Maple, а также двумерная и трехмерная графика, создаваемая графическими командами Maple, если задан режим вставки графики в рабочий лист, а не отображения ее в отдельном окне. В случае отображения графики на рабочем листе область вывода представляет собой *область вывода графики*.

Область ввода и вывода образуют *группу*, которая на рабочем листе отмечается квадратной скобкой. Отметим, что группа может содержать несколько областей ввода и, соответственно, вывода. Основное свойство группы заключается в том, что все ее операторы и команды выполняются за одно обращение к исполняющей системе Maple, т.е. нажатие клавиши <Enter> приводит к вычислению всех ее операторов и команд.

Область ввода

Область ввода — это область рабочего листа, в которой пользователь вводит информацию. Эта информация может быть двоякого рода: команды и операторы Maple или текстовые комментарии.

По умолчанию при создании нового рабочего листа (при загрузке Maple новый рабочий лист создается автоматически) устанавливается режим ввода команд и операторов. Указанием на это является приглашение ввода в строке рабочего листа — символ ' > ', сразу же после которого расположен мерцающий курсор. Пользователь вводит с клавиатуры команды и нажатием клавиши <Enter> передает их на обработку символьному анализатору Maple, который в зависимости от того, правильно или нет, они набраны, отображает в поле вывода либо результат выполнения команды, либо сообщение об ошибке.

Команды можно отображать либо в форме операторов Maple, либо в виде привычной математической записи. Например, для вычисления интеграла от функции $f(x)$ можно в поле ввода отобразить команду `int (f (x) , x)`, либо установить режим, при котором эта команда отобразится в привычной математической записи $\int f(x) dx$.

Можно задать режим, при котором в области ввода любая вводимая информация рассматривается как *текстовый комментарий* (Text). Такие комментарии удобны, когда необходимо разъяснить, что будет выполнять последующий оператор, или просто необходимо описать постановку задачи и основные моменты алгоритма ее решения. Система Maple широко используется по всему миру при обучении математике и техническим дисциплинам. Поэтому возможность вставки в рабочий лист текстовых комментариев, позволяющих создать учебный материал для изучения и демонстрации применения каких-то понятий, методов, алгоритмов и т.п., является полезным средством, облегчающим работу.

Пакет Maple позволяет изменять тип, размер и цвет шрифта, как в текстовых комментариях, так и в области ввода. Переключение между режимами ввода команд и текстового комментария осуществляется либо с помощью команд Text (Ввод текста) и Maple Input (Ввод команд) меню Insert (Вставить), либо с помощью кнопок на основной панели инструментов (кнопки с символами 'Σ' и 'T' в Classic Worksheet Maple; кнопки с символом 'Σ' в новом стандарте рабочего листа нет).

Если оператор не помещается в строке рабочего поля, он автоматически переносится на следующую строку. Пользователю нет необходимости следить за переносом длинных операторов или больших комментариев.

В одной строке можно вводить несколько операторов, разделенных точкой с запятой (;) или двоеточием (:), но иногда удобно задать несколько операторов по одному на строке, но так, чтобы при нажатии <Enter> они одновременно были переданы символному анализатору на выполнение. Это осуществляется нажатием комбинации клавиш <Shift> + <Enter> по завершении ввода оператора. В этом случае введенный оператор не пересылается на обработку интерпретатору, а система ожидает ввода пользователя, автоматически переместив курсор на новую строку, причем в новой строке приглашение на ввод не появляется. Таким способом можно ввести несколько операторов по одному на каждой строке, и после нажатия <Enter> все введенные операторы последовательно будут выполнены.

Область вывода

Результаты выполнения операторов, введенных в области ввода, отображаются в области вывода. Можно задать несколько форматов вывода — необходимый тип записи в области вывода задается командами Maple Notation (Maple запись), Character Notation (Символьная запись), Typeset Notation (Типографская запись) и Standart Math Notation (Стандартная математическая запись) подменю Preferences меню File — для Classic Worksheet Maple (для Maple — подменю Options меню Tools).

Обычно используется вывод в типографском стиле, так как он является наиболее читаемым.

Вывод графики

Как уже говорилось, вывод графики в Maple можно осуществлять непосредственно в рабочий лист (режим по умолчанию) или в отдельное окно. Задание соответствующих режимов Inline (Встроенная графика) и Window (Графика в окне) осуществляется для Maple в следующем порядке: меню Tools → подменю Options (Опции), после чего открывается диалоговое окно, в котором надо выбрать закладку Plotting (Отображение графики) и в разделе Plot Display выбрать Inline или Window. Для Classic Worksheet Maple — в такой последовательности File → Preferences, после чего открывается диалоговое окно, в котором надо выбрать закладку Plotting и в разделе Plot Display выбрать Inline или Window.

Вывод графики в окне полезен для отображения промежуточных результатов расчетов, так как графические изображения, расположенные на рабочем листе, значительно увеличивают размер файла при сохранении и требуют дополнительной оперативной памяти при открытии документа с графикой.

Работа с меню

Список доступных меню отображается в строке меню. При выборе какого-либо меню появляется панель с набором доступных команд и подменю данного меню. Команда отличается от подменю тем, что она сразу же выполняется при щелчке на ней левой кнопкой мыши, а подменю отображает свой список команд. Выбор какого-либо меню осуществляется щелчком мыши на имени этого меню.

На примере меню File поясним структуру раскрывающегося списка команд и подменю. При выборе меню появляется панель со списком команд и подменю.

В этой панели в левом столбце расположены имена (названия) команд, а справа расположены соответствующие каждой команде «быстрые клавиши» — комбинации клавиш, одновременное нажатие которых приводит к выполнению команды. Например, если на клавиатуре нажать одновременно клавиши <Ctrl> и <N> (такие действия принято обозначать <Ctrl> + <N>), то выполнится команда New меню File. Обратим внимание, что команды могут и не иметь быстрых клавиш, например команда Save As.

Если в правом столбце напротив имени команды отображается символ '▶', это означает, что данная команда является подменю, и при ее выборе отобразится дополнительная панель с командами (а может быть и подменю), например, команда – подменю Export As (Записать как).

Если в названии команды присутствует многоточие, это означает, что для выполнения данной команды требуется дополнительная информация, которую пользователь должен ввести в отображаемом диалоговом окне. Например, для команды Save As (Сохранить как) необходимо в диалоговом окне задать имя файла и его расположение (путь). Это можно сделать в диалоговом окне, отображаемом данной командой.

Обычно список команд разделен на логические группы, содержащие команды, выполняющие функционально-связанные действия. Такие группы отделяются в панели команд разделительной чертой. Например, для интерфейса Maple, команды меню File, связанные с печатью, выделены в одну группу.

Действия некоторых команд заключаются в установлении какого-либо режима работы системы Maple. Обычно они действуют как переключатели: включают или выключают соответствующий режим. Если режим, определенный командой, включен, то слева от ее имени отображается метка '✓' — «галочка».

Выбор меню и их команд можно производить и с помощью клавиатуры. Если посмотреть на названия меню, то можно заметить, что некоторые символы отображаются подчеркнутыми. Эти символы используются в качестве быстрых клавиш для выбора меню. При нажатии комбинации клавиш <Alt> + <подчеркнутый символ> отобразится список команд меню, в имени которого соответствующий символ подчеркнут. Например, для меню File таким символом является «F» (регистр клавиатуры не имеет значения).

Для выбора команд и подменю из списка меню можно применять «горячие клавиши». Они отличаются от быстрых клавиш тем, что простое нажатие на клавиатуре клавиши, соответствующей подчеркнутому символу в имени команды или подменю при раскрытом списке команд меню, приводит к выполнению соответствующей команды или отображению списка команд подменю. Например, нажатие клавиши <N> при раскрытом списке команд меню File приведет к выполнению команды New (Новый).

Если в списке команд раскрывающегося меню название команды отображается серым цветом, то данная команда не доступна. Обычно в Maple такое происходит с командами вставки содержимого буфера обмена или копирования в буфер обмена. Если он пуст, то, естественно, ничего нельзя вставить в документ, и поэтому команды вставки из буфера обмена не доступны. В случае если в рабочем листе нет выделенного фрагмента, то наоборот, ничего нельзя скопировать в буфер обмена, и соответствующие команды недоступны.

Виды меню

Интерфейс системы Maple является интерфейсом, реализующим работу с несколькими документами одновременно. Это означает, что в одном сеансе можно открыть несколько документов Maple и с помощью команд меню Window переключаться между ними или отобразить все окна открытых документов.

Концепция интерфейсов, поддерживающих работу со многими документами, так называемые *MDI-интерфейсы* (от английского Multiple Document Interface), предполагает, что при работе с разнородными документами список меню в строке меню может изменяться, чтобы предоставить пользователю команды, специфичные для того типа документа, с которым в данный момент он работает. В Maple существует два типа документов: рабочий лист и графический вывод (если задан режим вывода графики в отдельном окне). С каждым из них связаны свои списки меню, которые отображаются на строке меню при активизации соответствующего документа. Будем называть меню рабочего листа основным, а меню для графики — графическим.

Если графика вставлена в рабочий лист, то при помещении курсора в область графического вывода также отображается графическое меню, позволяющее выполнить ряд команд редактирования графического изображения. Отметим, что графические меню для двумерной и трехмерной графики отличаются, хотя некоторые названия меню совпадают, но списки команд в них разные.

Подробно графические меню будут рассмотрены в разделе, посвященном выводу графической информации. В данном разделе кратко остановимся на командах основного меню.

Основное меню

Основное меню отображается при работе с рабочими листами и состоит из следующих меню: File (Файл), Edit (Правка), View (Вид), Insert (Вставка), Format (Формат), Tools (Инструментарий), Window (Окно) и Help (Справка) для нового стандарта Maple. Для Classic Worksheet Maple — File, Edit, View, Insert, Format, Spreadsheet (Электронная таблица), Window и Help. Здесь кратко опишем основные, наиболее часто используемые при работе с Maple команды этих меню.

Команды меню File

Команды этого меню предназначены для создания, открытия, сохранения и печати рабочих листов.

Новый рабочий лист создается при помощи команды New (Новый). Отметим, что при запуске Maple автоматически создает новый рабочий лист. Для сохранения рабочего листа в файле существуют две команды: Save (Сохранить) и Save As (Сохранить как). Первая сохраняет рабочий лист в том же самом файле, из которого рабочий лист и был открыт, а вторая в файле с другим именем. Команда Save As отображает стандартное диалоговое окно системы Windows для задания имени файла и его расположения на жестком диске.

Открыть сохраненный ранее в файле рабочий лист можно командой Open (Открыть). При этом в отображаемом диалоговом окне следует задать имя и расположение файла. Команда Close (Закрыть) закрывает текущий активный рабочий лист, спрашивая пользователя о подтверждении сохранения файла перед закрытием, если в рабочий лист были внесены какие-то изменения.

Рабочие листы сохраняются в файлах с расширением mws. Существует возможность сохранить рабочий лист в файлах других форматов. Для этого необходимо в диалогической команде Save As указать соответствующий формат, либо воспользоваться командами подменю Export As (Экспортировать в формате). Мы не будем описывать эти другие форматы, только отметим, что система Maple позволяет сохранять рабочий лист в формате издательской системы LaTeX, достаточно широко используемой для написания математических текстов.

В этом же меню существуют команды печати Print (Печать) и предварительного просмотра Print Preview (Предварительный просмотр печати) рабочего листа, а также команда Print Setup (Настройка печати) для установки параметров печати.

Команды меню Edit

Команды этого меню позволяют редактировать и вносить правку в рабочий лист. В основном они предназначены для работы с текстовыми комментариями, чтобы придать им привлекательный и удобочитаемый вид.

Большинство команд этого меню аналогичны командам меню Edit (Правка) других приложений Windows. Команды Cut (Вырезать), Copy (Копировать) и Paste (Вставить) позволяют соответственно вырезать или скопировать выделенный текст в

Буфер обмена, а также вставить его содержимое в рабочий лист в месте расположения курсора.

Команда Undo (Отменить) позволяет отменить последнее действие, произведенное на рабочем листе. Команда Redo — наоборот, позволяет снова выполнить последнее действие, отмененное командой Undo.

Команда Select All (Выбрать весь лист) позволяет выделить все содержимое рабочего листа, например, для копирования в буфер обмена.

Команды меню View

Команды этого меню позволяют пользователю задать отображаемые элементы интерфейса пользователя системы Maple, а также отображение информации в рабочем листе.

Команды первой группы из списка команд являются простыми переключателями. Они включают или выключают режим отображения панели инструментов (команда Tool Bar), контекстной панели инструментов (команда Context Bar) и строки состояния (команда Status Bar). Если соответствующий элемент интерфейса отображается, то, как указывалось ранее, слева от имени команды стоит «галочка». Команда Show Invisible Characters (Показать непечатаемые символы) (для Classic Worksheet Maple) включает или выключает режим отображения непечатаемых символов на рабочем листе. Такими символами являются пробел и символ завершения строки ' ¶ '. Пробел при таком режиме отображается в виде точки, расположенной выше базовой линии строки (чтобы не спутать с точкой — знаком препинания, расположенной на базовой линии строки).

Команды подменю Zoom Factor (Изменить масштаб) позволяют изменять масштаб отображения рабочих листов: увеличивать или уменьшать размеры рабочих листов в окне системы Maple.

Команды подменю Bookmarks (Закладки) (для Classic Worksheet Maple) позволяют установить закладку в рабочем листе и быстро переходить к месту, отмеченному определенной закладкой. Это средство удобно для работы с большими документами, а также для отметки, например, текста, который требует редактирования и доработки.

Команды меню Window

Команды меню Window предназначены для работы с несколькими открытыми документами Maple. Они позволяют задать расположение окон документов в окне интерфейса Maple. Их можно расположить каскадом (команда Cascade), плиткой (команда Tile), горизонтально (команда Horizontal) и вертикально (команда Vertical).

С помощью команды Arrange Icons (Выстроить значки) (для Classic Worksheet Maple) упорядочиваются значки свернутых окон документов.

Две команды (для Classic Worksheet Maple) позволяют закрыть все открытые окна (Close All) или же закрыть окна, содержащие справочную информацию (Close All Help).

Кроме этих команд в меню Window отображаются все открытые в данной сессии документы, что позволяет легко и просто переходить от одного документа к другому.

Справочная система

Последним меню в списке является меню Help (Справка). Справка очень полезное и удобное средство. Она позволяет достаточно быстро найти нужную информацию, не прибегая к разнообразным книгам и пособиям, тем более, что в них обыкновенно отражена не вся информация о системе. Поэтому остановимся подробнее на получении справочной информации в системе аналитических вычислений Maple.

Организация справки

В основу справочной системы положено понятие гипертекста, т.е. документа, в котором имеются ссылки на другой документ. Таким образом, продвигаясь по этим ссылкам можно получать дополнительную информацию, уточняющую или дополняющую информацию, содержащуюся в исходном документе.

Это очень похоже на чтение книги, в которой имеются ссылки на другие главы, разделы или формулы той же книги, или даже на другую книгу. Только при чтении книги необходимо ее перелистывать и обращаться, например, к странице, на которую ссылаются в тексте, или вообще брать в библиотеке другую книгу.

В случае гипертекстовых документов не надо идти в библиотеку, а достаточно щелкнуть кнопкой мыши на соответствующей связи, и необходимый документ отобразится в окне интерфейса Maple.

Такие слова-ссылки (гиперссылки) в гипертекстовых документах отображаются другим цветом и могут быть подчеркнуты. В системе Справки Maple эти ссылки отображаются бирюзовым цветом и подчеркнуты. Если установить курсор мыши на такую связь и щелкнуть кнопкой мыши, то загрузится документ справочной системы, с которым установлена связь через слово-ссылку.

С помощью гиперссылок можно «путешествовать» по справочной системе Maple, добираясь до нужной информации. Такие «путешествия» полезны еще и тем, что можно натолкнуться на описание каких-то возможностей, с которыми пользователь еще не сталкивался, но которые могут оказаться полезными для него при решении задач проблемной области, в которой он работает.

Вызов справки с помощью меню Help

Команда Introduction (Введение) загружает в рабочее окно документ, содержащий оглавление справочной системы Maple. При этом изменяется список меню в полосе меню, панель инструментов, а контекстная панель инструментов становится недоступной. Документ Справки внешне представляет собой рабочий лист, который нельзя модифицировать, но который содержит гиперсвязи, позволяющие загружать другие документы Справки. Панель инструментов справочной системы содержит кнопки, ассоциированные с командами справочной системы.

В меню Edit справочной системы есть очень полезная команда — Copy Examples. Практически на каждой странице справки есть несколько примеров на использование описываемых функций. Эта команда позволяет скопировать в буфер обмена примеры, вставить их в рабочий лист и выполнить. Мы настоятельно рекомендуем проделывать эту процедуру для каждой, вновь изучаемой функции и анализировать получаемые результаты.

Команда Topic Search (Поиск раздела) (Topic Index — для Maple) меню Help позволяет найти необходимый раздел справки, а команда Full Text Search (Поиск текста) (Search — для Maple) позволяет найти все страницы Справки, в которых найден заданный в диалоговом окне этой команды текст.

Полезная команда History (История) отображает все просмотренные в данном сеансе страницы Справки и позволяет перейти на любую из них.

При работе со Справкой может быть открыто много справочных страниц. Чтобы закрыть все окна страниц Справки, необходимо в меню Window выбрать команду Close All Help (Закреть все страницы Справки).

Вызов справки из рабочего листа

Получать справочную информацию можно и не используя команды меню Help, набрав команду ?topic, где topic означает имя пакета (библиотеки), команды или служебного слова Maple. (Заметим, что эта команда не завершается точкой с запятой или двоеточием.) Для получения информации о команде subtopic из пакета topic можно воспользоваться следующими эквивалентными формами команды получения справки: ?topic, subtopic; и ?topic[subtopic];. Обратим внимание на то, что эти команды завершаются разделителем (;).

Вместо команд со знаком вопроса в начале можно использовать следующие формы команды help, завершающиеся точкой с запятой:

```
help(topic);  
help(topic, subtopic);  
help(topic[subtopic]);
```

Все перечисленные вызовы справки требуют задания команд в области ввода, что может оказаться неудобным. Система Maple предоставляет более удобный способ получения справки по команде, заданной в рабочем листе. Для этого необходимо поместить курсор на имени команды или служебного слова, и нажать клавишу <F1>. Такой способ, конечно, удобен, если необходимо быстро получить информацию, например, о параметрах нужной команды или об альтернативных формах ее задания. При этом, естественно, необходимо знать имя команды, поэтому предыдущие способы или обращение к командам меню Help остаются единственными способами для определения, например, всех команд какого-нибудь пакета.

Основы работы

Система аналитических вычислений Maple является интерактивной системой. Это означает, что пользователь вводит команды в рабочем листе и, нажав клавишу <Enter>, сразу же передает ее аналитическому анализатору системы, который выполняет ее. Если команда введена правильно, то в области вывода появляется результат ее выполнения, если она содержит синтаксические ошибки, система печатает сообщение об этом. Для исправления ошибки следует вернуться к оператору, откорректировать его и снова выполнить. После выполнения введенной команды система ожидает очередной команды от пользователя.

Каждый оператор или команда должны *обязательно* завершаться разделительным знаком. Таких разделительных знака в Maple два — точка с запятой (;) и двоеточие

(:). Если предложение завершается точкой с запятой, то оно вычисляется, а в области вывода отображается результат. При использовании двоеточия в качестве разделителя команда выполняется, но результаты ее работы не отображаются в области вывода рабочего листа. Это удобно при программировании в Maple, когда нет необходимости в выводе каких-то промежуточных результатов, получаемых из операторов цикла, так как вывод этих результатов занял бы много места на рабочем листе и время выполнения существенно увеличилось бы.

Объекты, переменные и выражения

Как и в любой интерактивной системе, в Maple реализован свой язык, с помощью которого происходит общение пользователя с системой. Базовыми понятиями являются объекты и переменные.

Простейшими *объектами*, с которыми может работать Maple, являются числа, константы и строки. Если читатель знаком с каким-либо языком программирования, то эти объекты соответствуют литерным (буквенным) константам разных типов данных в языках программирования.

Числа могут быть целыми, обыкновенными дробями, радикалами и числами с плавающей точкой. Первые три типа чисел позволяют выполнять точные вычисления разнообразных математических выражений, без округлений. Числа с плавающей точкой являются приближенными, в которых ограничено число значащих цифр.

Целые числа задаются в виде последовательности цифр от 0 до 9. Отрицательные числа задаются со знаком минус (-) перед числом. *Дробь* задается с помощью операции деления двух *целых* чисел — для задания дробей используется символ деления (/). Заметим, что Maple автоматически производит операцию сокращения дробей. *Радикалы* задаются как целые или дробные числа, возведенные в дробную степень (для возведения в степень используются символы (^) или (**)), или с помощью функции sqrt() с целым или дробным параметром. Здесь также сразу производятся возможные упрощения. Для задания радикалов имеется и специальная функция — surd(). Например,

> 2^(1/5);	$2^{(1/5)}$
> 3**(1/2);	$\sqrt{3}$
> sqrt(7);	$\sqrt{7}$
> surd(5,3);	$5^{(1/3)}$

Вычисления с целыми, дробями и радикалами являются *абсолютно точными*, так как при работе с этими типами чисел Maple не происходит никаких округлений в отличие чисел с плавающей точкой.

Числа с плавающей точкой задаются в виде целой и дробной частей, разделенных десятичной точкой, и предшествующим знаком числа, например 2.3456, -3.415. Можно задать целый показатель степени после букв 'e' или 'E', в которую необходимо возвести число 10 и потом умножить его на заданную мантиссу числа. Например, 2.345e4 соответствует числу 23450.0. Таким образом, можно представлять очень большие по абсолютному значению числа (показатель степени положительное число) или очень маленькие (показатель степени отрицательное число).

Из чисел можно составлять математические выражения с помощью арифметических операций.

Символы арифметических операций приведены в таблице 1.

Таблица 1. Символы арифметических операций

Символ	Операция
+	Сложение
-	Вычитание
*	Умножение
/	Деление
^ или **	Возведение в степень
!	Факториал (применим только к натуральным числам)

Последовательность выполнения арифметических операций соответствует стандартным правилам старшинства операций в математике: сначала производится возведение в степень, затем умножение и деление, а в конце — сложение и вычитание. Все действия выполняются слева направо. Операция вычисления факториала имеет наибольший приоритет. Для изменения последовательности арифметических операций следует использовать круглые скобки.

Если в выражении все числа являются целыми, дробями или радикалами, то результат представляется также с использованием этих типов данных, но если в выражении присутствует число с плавающей точкой, то там, где это возможно, Maple будет производить вычисления с абсолютной точностью, но некоторые подвыражения будут вычислены с плавающей точкой, например,

$$\begin{aligned}
 &> 4.0 + \sqrt{2/3} * 3.0^{(1/3)} / \sqrt{2}; \\
 &4.0 + 0.2403749283 \sqrt{6} \sqrt{2}
 \end{aligned}$$

Скажем так, Maple всегда пытается произвести вычисления с абсолютной точностью. Если это не получается, тогда он подключает арифметику с вещественными числами.

Система Maple умеет работать с комплексными числами. Для мнимой единицы $\sqrt{-1}$ в Maple используется зарезервированная константа I. Задание комплексного числа в Maple не отличается от обычного задания его в математике. Maple умеет выполнять все арифметические действия над комплексными числами. Для выделения из комплексного числа действительной и мнимой части предусмотрены две функции Re() и Im(). Вычислить аргумент комплексного числа можно с помощью функции

`argument()`, а построить комплексно сопряженное число — с помощью функции `conjugate()`.

Кроме чисел, задаваемых пользователем, Maple содержит целый ряд предопределенных именованных констант — констант, к значению которых можно обращаться с помощью некоторого имени. Часть этих констант не может быть изменена, а часть можно изменять.

К неизменяемым константам относятся следующие:

Таблица 2. Неизменяемые константы

Константа	Значение
<code>false</code>	Значение «ложь» при работе с булевыми переменными
<code>true</code>	Значение «истина» при работе с булевыми переменными
<code>gamma</code>	Константа Эйлера $\gamma = \lim_{n \rightarrow \infty} \left(\sum_{i=1}^n 1/i - \ln(n) \right) \approx 0.57721566\dots$
<code>Pi</code>	Число $\pi \approx 3.14159265\dots$
<code>I</code>	Мнимая единица $\sqrt{-1}$
<code>infinity</code>	Бесконечность ∞
<code>Catalan</code>	Константа, являющаяся суммой ряда $\sum_{i=1}^n (-1)^i / (2i+1)^2 \approx 0.9159655942\dots$

Константы, значения которых могут быть переопределены, — это константы, задающие необходимые для работы системы значения. К наиболее важным можно отнести две константы, позволяющие влиять на точность вычислений: `Digits` и `Order`. Первая константа задает число значащих цифр для операций с числами с плавающей точкой. По умолчанию она имеет значение 10. Константа `Order` определяет количество членов в разложении функции в ряд (по умолчанию установлена равной 6).

Кроме чисел Maple позволяет работать со *строкой* — любым набором символов, заключенным в обратные кавычки, например,

```
> `Это пример строки в Maple`;
```

Это пример строки в Maple

Для работы со строками и строковыми переменными существует набор функций, с которым можно ознакомиться по справочной системе. Например, соединение строк можно осуществить с помощью операции конкатенации (`||`), или обращением к функции `cat()`:

```
> `Map`||`le 9`;
```

Maple 9

```
> cat(`Mathe`, `matica`);
```

Mathematica

Каждая *переменная* Maple имеет имя — последовательность символов, начинающаяся с буквы, причем строчные и прописные буквы считаются различными. Про такие системы говорят, что они чувствительны к регистру. Кроме букв в именах переменных могут использоваться также цифры и знак подчеркивания. Примеры *различных* имен: `MyName`, `myname`, `my_name`.

В качестве имен запрещено использовать зарезервированные слова языка Maple: `and`, `break`, `by`, `catch`, `description`, `do`, `done`, `elif`, `else`, `end`, `error`, `export`, `fi`, `finally`, `for`, `from`, `global`, `if`, `in`, `intersect`, `local`, `minus`, `mod`, `module`, `next`, `not`, `od`, `option`, `options`, `or`, `proc`, `quit`, `read`, `return`, `save`, `stop`, `then`, `to`, `try`, `union`, `use`, `while`.

Важной операцией в Maple является операция присваивания (`:=`). Синтаксис этого оператора следующий:

```
переменная := выражение;
```

Здесь в левой части задается имя переменной, а в правой части любое выражение, которое может быть числовым, символьным или просто другой переменной. Семантика (смысл) этого оператора в том, что переменной в левой части присваивается значение выражения, стоящего в правой части.

Переменные позволяют хранить и обрабатывать разнообразные типы данных, с которыми работает Maple. Мы уже знакомы с такими типами данных, как целый (`integer`), дробь (`fraction`), вещественный (`float`) и строка (`string`). Кроме этих типов данных существует еще большое множество типов, необходимых для выполнения аналитических преобразований: функция (`function`), индексные данные (`indexed`), множество (`set`), список (`list`), разложение (`series`), последовательность выражений (`exprseq`) и другие.

По умолчанию переменная Maple имеет тип `symbol`, и ее значением является ее собственное имя. Поэтому простое объявление переменной `g` оператором `g`; приведет к отображению в области вывода рабочего листа имени этой переменной

```
> g;
```

g

Функция `whattype()` определяет тип выражения или переменной, заданной в качестве ее параметра:

```
> whattype(g);
```

symbol

То, что переменная по умолчанию является символьной, оказывается очень полезным при задании функций. Если имя функции Maple задано не совсем правильно, или такой функции не существует, или не подключен пакет, где она расположена, то ответом Maple на попытку вычислить ее будет отображение в области вывода не результата выполнения функции, а полностью повторенная строка области ввода.

При присвоении переменной какого-нибудь значения, ее тип изменяется на тип присвоенного ей значения.

Переменные можно использовать для составления выражений наряду с числами. Все, сказанное выше о числовых выражениях и порядке их вычисления, относится и к выражениям, содержащим переменные.

Обычно в математических выражениях используются разнообразные математические функции. В Maple изначально определен большой набор стандартных математических функций, начиная от элементарных и заканчивая специальными функциями, используемыми при решении сложных задач математической физики. Справку о всех, имеющихся в Maple функций, можно получить, выполнив команду `?inifunction`.

В таблице 3 представлены основные математические функции и соответствующий им синтаксис Maple:

Таблица 3. Основные математические функции

	Функция	Синтаксис Maple
e^x		<code>exp(x)</code>
$\ln x$		<code>ln(x)</code> или <code>log(x)</code>
$\log_{10} x$		<code>log10(x)</code>
$\log_a x$		<code>log[a](x)</code>
\sqrt{x}		<code>sqrt(x)</code>
$ x $		<code>abs(x)</code>
$\operatorname{sgn}(x)$		<code>signum(x)</code>
$n!$		<code>n!</code>

Тригонометрические и гиперболические функции представлены в таблице 4. Обратим внимание на несоответствие записи некоторых функций в русскоязычной и в англоязычной математической литературе, например функция тангенса. Значения параметров тригонометрических функций задаются в радианах.

Таблица 4. Тригонометрические и гиперболические функции

	Функция	Синтаксис Maple
$\sin x$		<code>sin(x)</code>

$\cos x$	$\cos(x)$
$\operatorname{tg} x$	$\tan(x)$
$\sec x$	$\sec(x)$
$\operatorname{cosec} x$	$\csc(x)$
$\operatorname{ctg} x$	$\cot(x)$
$\operatorname{sh} x$	$\sinh(x)$
$\operatorname{ch} x$	$\cosh(x)$
$\operatorname{th} x$	$\tanh(x)$
$\operatorname{sech} x$	$\operatorname{sech}(x)$
$\operatorname{cosech} x$	$\operatorname{csch}(x)$
$\operatorname{cth} x$	$\operatorname{coth}(x)$

Обратные тригонометрические и гиперболические функции представлены в таблице 5.

Таблица 5. Обратные тригонометрические и гиперболические функции

Функция	Синтаксис Maple
$\arcsin x$	$\arcsin(x)$
$\arccos x$	$\arccos(x)$
$\operatorname{arctg} x$	$\arctan(x)$
$\operatorname{arcsec} x$	$\operatorname{arcsec}(x)$
$\operatorname{arccosec} x$	$\operatorname{arccsc}(x)$
$\operatorname{arcctg} x$	$\operatorname{arccot}(x)$
$\operatorname{arcsh} x$	$\operatorname{arcsinh}(x)$
$\operatorname{arcch} x$	$\operatorname{arccosh}(x)$
$\operatorname{arth} x$	$\operatorname{arctanh}(x)$
$\operatorname{arcsech} x$	$\operatorname{arcsech}(x)$
$\operatorname{arccosech} x$	$\operatorname{arccsch}(x)$
$\operatorname{arccth} x$	$\operatorname{arccoth}(x)$

Приведем еще в таблице 6 некоторые специальные функции, используемые в математической физике.

Таблица 6. Некоторые специальные функции

Функция	Синтаксис Maple
----------------	------------------------

Функция Бесселя первого рода $J_\nu(x)$ `BesselJ(nu, x)`

Функция Бесселя второго рода $Y_\nu(x)$ — `BesselY(nu, x)`
функция Вебера

Модифицированная функция Бесселя первого рода $I_\nu(x)$ `BesselI(nu, x)`

Модифицированная функция Бесселя второго рода $K_\nu(x)$ — функция Макдональда `BesselK(nu, x)`

Интеграл вероятности или функция `erf(x)`

$$\text{ошибок } \Phi(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-s^2} ds$$

Дополнительный интеграл вероятности `erfc(x)`

$$1 - \Phi(x) = \frac{2}{\sqrt{\pi}} \int_x^\infty e^{-s^2} ds$$

Дельта-функция Дирака `Dirac(x)`

Единичная функция Хевисайда `Heaviside(x)`

Другие специальные функции можно найти в Справке.

Команды

Технология работы в Maple заключается в том, что пользователь создает переменные и выражения и производит над ними некоторые действия в соответствии с алгоритмом решения поставленной задачи, используя стандартные функции или написанные собственные процедуры.

Синтаксис вызова стандартной команды или функции следующий:

`команда (пар_1, пар_2, ... , пар_n) ;`

или

`команда (пар_1, пар_2, ... , пар_n) :`

Здесь команда — это имя вызываемой функции, а `пар_1, пар_2, ...` означают необходимые для выполнения команды параметры, которые могут быть переменными или выражениями, причем их тип должен соответствовать типу параметров соответствующей функции. Первая форма задания команды (с завершающей точкой с запятой) осуществляет отображение результатов ее выполнения в области вывода, тогда как при второй форме (с завершающим двоеточием) команда выполняется, но никакого вывода результатов не происходит.

Система Maple является интуитивной системой и обычно имя функции соответствует действию, которое она выполняет (следует учитывать, что все имена заданы на английском языке). Например, ясно, что функция с именем `simplify()`

осуществляет некоторые упрощения над выражением, заданным в качестве ее параметра.

Для некоторых команд существуют две формы обращения: активная и пассивная. Имя команды в первом случае (активная форма) начинается со строчной буквы. При этом команда немедленно будет выполнена. Пассивная форма команды не выполняется немедленно ядром Maple, а просто в области вывода отображает математическую запись (формулу) того, что эта команда может сделать. Ее имя начинается с прописной буквы. В дальнейшем эту команду можно выполнить с помощью команды `value()`.

Команды и функции, являющиеся частью ядра системы Maple, всегда доступны пользователю, тогда как для вызова других команд и функций необходимо подключить библиотеку или пакет, в которых они расположены. Для этого используются команды `readlib()` и `with()`. Первая подключает библиотеку, вторая — пакет. Параметром этих команд является имя библиотеки или пакета, функции которых пользователь желает использовать.

Команды и функции ядра

Наиболее часто используемые при аналитических преобразованиях команды и функции Maple располагаются в его системном ядре — части программного обеспечения системы аналитических вычислений, постоянно находящейся в памяти компьютера. К таким командам относятся команды, выполняющие разнообразные преобразования выражений, получающие решение уравнений и систем уравнений, дифференцирующие функции и т.д. В данном разделе мы рассмотрим команды, наиболее часто используемые при выполнении аналитических вычислений.

Команда `simplify()`

Начнем с команды упрощения выражений — команды `simplify()`. Эта команда предназначена для упрощения разнообразных выражений, включающих рациональные дроби (алгебраические выражения), содержащих тригонометрические и обратные тригонометрические функции, логарифмы и экспоненты, то есть с ее помощью можно попытаться упростить заданное выражение. Почему попытаться? Дело в том, что Maple может его упростить, а может и не упростить, так как он использует свои внутренние алгоритмы упрощения, которые могут содержать не все возможные формулы, используемые человеком при упрощении выражений. Кроме того, очевидно, само понятие простого выражения не однозначно.

Синтаксис команды упрощения следующий:

```
simplify(выражение);
```

В скобках в качестве параметра передается выражение, подлежащее упрощению. Команда `simplify()` ищет в выражении вызовы функций, квадратные корни, радикалы и степени и инициирует подходящие процедуры упрощения. Maple содержит большое число подобных процедур, относящихся к упрощению выражений, содержащих определенный набор элементарных и специальных математических функций. Мы перечислим часть из них, остальные можно найти в справке по этой команде (например, установив курсор в рабочем листе на ее имя и нажав <F1>): `exp` — экспоненты, `ln` — логарифмы, `sqrt` — квадратные корни,

`trig` — тригонометрические функции, `radical` — радикалы (дробные степени), `power` — степени, экспоненты, логарифмы, `BesselJ` — функции Бесселя и многие другие.

В вызове команды можно задать конкретные процедуры упрощения, и тогда только они будут использоваться для упрощения заданного выражения, а не весь возможный, установленный по умолчанию набор. Такой вызов обеспечивается следующим синтаксисом команды:

```
simplify(выражение, n1, n2, ...);
```

Здесь `n1`, `n2` и т.д. являются именами процедур упрощения: `Ei`, `GAMMA`, `RootOf`, `@`, `hypergeom`, `ln`, `polar`, `power`, `radical`, `siderel`, `sqrt`, `trig`. Полную информацию о формулах упрощения при использовании перечисленных значений параметров можно получить с помощью команды `?simplify[имя]`, где `имя` — одно из значений параметров функции упрощения.

При упрощении выражения можно предположить, что все переменные в нем являются, например, положительными, или принадлежат некоторому отрезку действительных чисел. Это осуществляется заданием ключевого параметра `assume=свойство`. Форма вызова команды в этом случае имеет вид:

```
simplify(выражение, assume=свойство);
```

где значение ключевого параметра `свойство` может принимать одно из следующих значений: `complex` — комплексная область, `real` — действительная область, `positive` — положительные действительные числа, `integer` — целые числа, `RealRange(a,b)` — интервал (a, b) действительных чисел.

Отметим, что всегда при упрощении Maple предполагает, что там, где это возможно, переменные изменяются в области комплексных чисел. Это надо учитывать, и, если область изменения переменных сужается, это надо дополнительно сообщить системе Maple, например, с помощью оператора `assume`. В противном случае невозможно будет упростить выражение. Например,

```
> g:=sqrt(x^2);
```

$$g := \sqrt{x^2}$$

```
> simplify(g);
```

$$\operatorname{csgn}(x) x$$

```
> simplify(g, assume=real);
```

$$|x|$$

```
> simplify(g, assume=positive);
```

$$x$$

```
> simplify(g, symbolic);
```

$$x$$

Команда `expand()`

Основное назначение команды `expand()` — представить произведение в виде суммы, то есть данная команда раскрывает скобки в алгебраическом выражении. Она выполняется для любого полинома. Для частного двух полиномов (рациональная алгебраическая дробь) она раскрывает скобки в числителе и делит каждый член полученного выражения на знаменатель, с которым она не производит никаких преобразований.

Кроме того, данная команда умеет работать с большинством математических функций и знает, как раскрывать скобки в выражениях, содержащих тригонометрические и гиперболические функции, логарифмы, специальные математические функции и другие.

Синтаксис этой команды следующий:

```
expand(выр, выр1, выр2, ... , вырn);
```

где `выр` является выражением, в котором необходимо раскрыть скобки, необязательные параметры `выр1`, `выр2`, ... , `вырn` указывают системе, что в данных выражениях в заданном преобразуемом выражении `выр` раскрывать скобки не надо. Например,

```
> expand((x+1)*(y+z));
```

$$x \sim y + x \sim z + y + z$$

```
> expand((x+1)*(y+z), x+1);
```

$$(x \sim + 1)y + (x \sim + 1)z$$

```
> expand(BesselJ(2, z));
```

$$\frac{2 \text{BesselJ}(1, z)}{z} - \text{BesselJ}(0, z)$$

Команда `factor()`

Назначение команды `factor()` — разложить на множители полином от нескольких переменных. Под полиномом в Maple понимается выражение, содержащее неизвестные величины, в котором каждый член представлен в виде произведения целых неотрицательных степеней неизвестных величин с некоторыми числовыми или алгебраическими коэффициентами. Неизвестные величины в полиноме могут быть сами функциями от неизвестных величин. Следует иметь в виду, что эта команда раскладывает многочлен на множители над тем числовым полем, которому принадлежат коэффициенты полинома. При обращении к этой команде вторым, необязательным, параметром может выступать явное указание, над каким числовым полем следует осуществить разложение полинома на множители. Например,

```
> factor(x^3+5);
```

$$x^3 + 5$$

```
> factor(x^3+5, 5^(1/3));
```

$$(x^2 - x 5^{(1/3)} + 5^{(2/3)})(x + 5^{(1/3)})$$

```

> factor(x^3+5, {5^(1/3), (-3)^(1/2)});
      (-2 x + 5^(1/3) + 5^(1/3) sqrt(-3)) (2 x - 5^(1/3) + 5^(1/3) sqrt(-3)) (x + 5^(1/3))
      -----
      4
> factor(x^3+5.0);
      (x + 1.709975947) (x^2 - 1.709975947 x + 2.924017740)
> factor(x^3+5, complex);
      (x + 1.709975947) (x - 0.8549879733 + 1.480882610 I)
      (x - 0.8549879733 - 1.480882610 I)

```

Команда combine()

Команда `combine()` приводит несколько членов в выражении, представленном суммой, произведением или степенями, к одному члену. Эта команда в некотором смысле противоположна команде `expand()`. Например,

```

> a:=4*sin(x)^3+exp(y)*exp(x);
      a := 4 sin(x)^3 + e^y e^x
> combine(a);
      -sin(3 x) + 3 sin(x) + e^(y+x)
> combine(a, trig);
      -sin(3 x) + 3 sin(x) + e^y e^x
> combine(a, exp);
      4 sin(x)^3 + e^(y+x)

```

Команда `combine` «знает» практически все правила преобразования элементарных функций. Если вторым ее параметром задать одно из имен, которое соответствует используемым в Maple функциям, то будут применяться только правила преобразования соответствующих функций. Для функций, правила преобразования которых зависят от значения их аргументов, или которые имеют ограничения на значения аргументов, можно задать третий параметр `symbolic`, который будет предписывать функции `combine` не обращать внимания на интервалы изменения аргументов подобных функций, а осуществлять формальные символичные преобразования в соответствии с формулами преобразования этих функций.

Более подробную информацию об опциях команды `combine` можно получить, выполнив команду `?combine` или `?combine[опция]`.

Команда collect()

Команда `collect()` работает с обобщенными полиномами нескольких переменных, то есть с полиномами, в которых в качестве неизвестных могут выступать функции с аргументами, являющимися неизвестными величинами Maple.

Синтаксис команды имеет три формы

```
collect (expr, x);  
collect (expr, x, form, func);  
collect (expr, x, func);
```

Здесь параметр *x* представляет имя неизвестной величины, список или множество неизвестных в случае полинома от нескольких переменных или имя функции с аргументом-неизвестной в выражении, представленном первым параметром *expr*, и относительно степени которого осуществляется приведение коэффициентов.

Параметр *form* имеет смысл для полиномов от нескольких переменных и определяет алгоритм приведения подобных членов, причем неизвестные, при степенях которых приводятся подобные члены, должны быть заданы в виде списка или множества. Этот параметр имеет два значения *recursive* и *distributed*. По умолчанию используется значение *recursive*. В этом случае приводятся подобные члены при степенях первой неизвестной в списке, затем в полученных коэффициентах приводятся подобные члены относительно степеней второй неизвестной в списке и т.д. Значение *distributed* указывает на приведение коэффициентов при членах, содержащих всевозможные произведения степеней неизвестных в списке или множестве, причем суммарная степень всех переменных возрастает от наименьшей к наибольшей.

Параметр *func* определяет имя команды, которая применяется к полученным в результате коэффициентам при соответствующих степенях неизвестных.

Например,

```
> f := a*ln(x) - ln(x)*x - x;  
f := a ln(x) - ln(x) x - x  
> collect(f, ln(x));  
(a - x) ln(x) - x  
> p := x*y + a*x*y + y*x^2 - a*y*x^2 + x + a*x;  
collect(p, [x, y], recursive);  
p := x y + a x y + y x^2 - a y x^2 + x + a x  
(1 - a) y x^2 + ((1 + a) y + 1 + a) x  
> collect(p, [y, x], recursive);  
((1 - a) x^2 + (1 + a) x) y + (1 + a) x  
> collect(p, [x, y], distributed);  
(1 + a) x + (1 + a) x y + (1 - a) y x^2  
> collect(p, x, factor);  
-(-1 + a) y x^2 + (1 + a) (y + 1) x
```

Команда `normal()`

Назначение команды `normal()` — упростить алгебраическую дробь, сократив и числитель, и знаменатель на наибольший общий делитель. Она имеет две формы вызова:

```
normal(f);  
normal(f, expanded);
```

где f — алгебраическая дробь, а параметр `expanded` служит для указания того, что после сокращения дроби в числителе и знаменателе раскрываются скобки.

Если параметр f задан в виде списка, множества, последовательности, ряда, уравнения, отношения или функции, то команда `normal()` последовательно применяется к компонентам f . Например, в случае ряда, это означает, что упрощаются коэффициенты ряда.

Команда `assume()`

Команда `assume()` накладывает ограничения на неизвестные величины Maple. Она имеет следующий синтаксис

```
assume(x, prop);
```

Здесь x представляет любую неопределенную переменную Maple или выражение с такими переменными, а параметр `prop` может принимать значения, равные названиям свойств — специальным символьным именам, зарезервированным системой Maple для задания различных ограничений. Некоторые из наиболее употребительных свойств перечислены в таблице 7.

Таблица 7. Свойства числовых переменных и выражений

Название свойства	Описание
<code>negative</code>	Отрицательные вещественные числа из интервала $(-\infty, 0)$
<code>nonnegative</code>	Неотрицательные вещественные числа из интервала $[0, +\infty)$
<code>positive</code>	Положительные вещественные числа из интервала $(0, +\infty)$
<code>natural</code>	Натуральные числа
<code>posint</code>	Целые положительные числа
<code>odd</code>	Нечетные числа
<code>even</code>	Четные числа
<code>complex</code>	Комплексные числа
<code>real</code>	Вещественные числа

rational	Рациональные числа
irrational	Иррациональные числа
integer	Целые числа

Познакомиться со всем перечнем используемых в Maple свойств и их имен можно на странице Справки, которая отображается командой `?property`.

Команда `assume()` может получать несколько пар (x, prop) или математических отношений в качестве своих параметров. В этом случае все заданные ограничения действуют одновременно. Так, наложение ограничений

```
assume(x>-1, x<1);
```

соответствует тому, что переменная $x \in (-1, 1)$.

Новое ограничение, накладываемое новой командой `assume()` на ту же переменную, отменяет предыдущее. Если по ходу решения задачи необходимо постепенно добавлять ограничения на переменную, то можно использовать команду `additionally()`, параметры которой полностью соответствуют параметрам команды `assume()`.

Чтобы снять все ограничения для какой-то переменной x , следует просто выполнить присваивание

```
x := 'x';
```

то есть этой переменной присвоить ее же символьное имя (имя переменной, заключенное в одинарные кавычки).

С помощью функции `is()` можно определить, удовлетворяет ли некоторая переменная рабочего листа определенному свойству. Эта функция возвращает значение `true`, если все возможные значения переменной соответствуют заданному свойству, и значение `false` — в противном случае.

Команда `about()` отображает информацию о наложенных ограничениях на неизвестную величину.

Команда `subs()`

При выполнении математических преобразований часто необходимо произвести замену переменных в выражении, функции, уравнении и т.д., то есть вместо какой-то переменной подставить ее представление через некоторые другие переменные. Для этих целей в системе Maple существует команда `subs()`, синтаксис которой следующий:

```
subs( старое_выражение = новое_выражение, выражение );
subs( s1, ... , sn, выражение );
```

где каждое из s_1, \dots, s_n является уравнением или множеством/списком уравнений. (Уравнение — два выражения, соединенные знаком равенства (=).)

Первая форма команды анализирует выражение, выделяет в нем все вхождения `старое_выражение` и подставляет вместо них `новое_выражение`. Вторая

форма позволяет выполнить серию подстановок в выражение. Подстановки выполняются последовательно, начиная с $s1$. Это означает, что после выполнения первой подстановки, определенной уравнением $s1$, Maple отыскивает вхождения левой части уравнения $s2$ во вновь полученном выражении и заменяет каждое такое вхождение выражением, заданным в левой части уравнения $s2$. Если подстановки заданы в виде списка или множества уравнений, то они выполняются одновременно, то есть вхождения выражений, заданных в левых частях уравнений, определяются в исходном параметре выражение.

Здесь следует немного сказать о внутреннем представлении объектов Maple. Каждый объект Maple, в том числе и выражение, делится на подобъекты первого уровня, которые в свою очередь, также делятся на подобъекты и т.д. Этот процесс продолжается до тех пор, пока не дойдем до базисных элементов Maple (объектов основных типов: целые, вещественные, дроби и т.д.). В результате получается ветвящаяся, древоподобная структура представления любого объекта Maple через базовые элементы. При работе с большими выражениями часто требуется извлекать отдельные элементы структуры и преобразовывать их.

Команда `nops(объект)` определяет количество объектов первого уровня, а команда `op(объект)` выдает их в виде последовательности выражений. Эта же команда позволяет извлечь подобъекты первого уровня, указав в качестве первого параметра порядковый номер подобъекта первого уровня. Например,

```
> s:=C[1]*sin(sqrt(lambda)*x)+C[2]*cos(sqrt(lambda)*x);
```

$$s := C_1 \sin(\sqrt{\lambda} x) + C_2 \cos(\sqrt{\lambda} x)$$

```
> nops(s);
```

2

```
> s1:=op(s);
```

$$s1 := C_1 \sin(\sqrt{\lambda} x), C_2 \cos(\sqrt{\lambda} x)$$

Здесь переменная s содержит в себе два объекта первого уровня. Список этих объектов представлен в переменной $s1$.

```
> whattype(s); whattype(s1);
```

+

exprseq

Получить саму операцию, операнды которой нам теперь известны, можно обращением к той же самой функции `op()`, но с двумя параметрами — первый должен равняться нулю, а второй ссылаться на выражение:

```
> op(0,s);
```

+

Таким образом, переменная s представляет собой сумму, а переменная $s1$ — последовательность. Извлекаем из выражения s подобъекты первого уровня, присваивая им имена $s11$ и $s12$:

```
> s11:=op(1,s);s12:=op(2,s);
      s11 := C1 sin( $\sqrt{\lambda} x$ )
      s12 := C2 cos( $\sqrt{\lambda} x$ )
```

Рассмотрим теперь выражение s11:

```
> nops(s11);
      2
> s11_1:=op(1,s11);s11_2:=op(2,op(1,s));
      s11_1 := C1
      s11_2 := sin( $\sqrt{\lambda} x$ )
```

Выражение s11 содержит в себе два объекта первого уровня, которым мы присвоили имена s11_1 и s11_2.

```
> whattype(s11_1);nops(s11_1);op(0,s11_1);
      indexed
      1
      C
```

Таким образом, переменная s11_1 есть индексированная переменная C.

```
> whattype(s11_2);nops(s11_2);op(0,s11_2);
      function
      1
      sin
```

Переменная s11_2 представляет собой функцию. Можно продолжить исследование объекта s11_2:

```
> nops(s11_2);op(s11_2);
      1
       $\sqrt{\lambda} x$ 
> nops(%);
      2
> op(1,%);op(2,%%);
       $\sqrt{\lambda}$ 
      x
```

Здесь мы воспользовались специальными переменными %, %% и %%%, хранящими результат выполнения предыдущей, пред-предыдущей и пред-пред-предыдущей

команды. Эти специальные переменные можно использовать, если мы не хотим сохранять результат выполнения команды.

Теперь можно вернуться к описанию команды подстановки `subs()`. Так вот, эта команда заменяет выражение, заданное в левой части одного из первых параметров, на выражение из правой части, только если это выражение из левой части совпадает с одним из подбъектов в структурном представлении выражения, в котором производится подстановка. Такая подстановка называется «синтаксической подстановкой». Поэтому, например, подстановка в выражение x^3 вместо x^2 выражения $1 - y^2$ с помощью команды `subs(x^2=1-y^2, x^3)` не приведет к желаемому результату. Однако в Maple имеется специальная команда

```
algsubs( старое_выражение = новое_выражение, выражение );
```

предназначенная для алгебраической подстановки и которая выполнит не решаемую командой `subs` вышеуказанную подстановку:

```
> algsubs(x^2=1-y^2, x^3);  
      (1 - y^2) x
```

Если известно, какой операнд выражения необходимо заменить, то можно использовать еще одну полезную команду `subsop()` со следующим синтаксисом

```
subsop(eq1, eq2, ..., eqn, expr);
```

Параметры `eq1, eq2, ..., eqn` представляют собой уравнения, в левых частях которых стоят порядковые номера операндов выражения, заданного последним параметром `expr`, а правые части представляют выражения, на которые заменяются соответствующие операнды. Например,

```
> p := x^7+8*x^6+x^2-9;  
      p := x^7 + 8 x^6 + x^2 - 9  
  
> op(2, p);  
      8 x^6  
  
> subsop( 2=y, p );  
      x^7 + y + x^2 - 9
```

Подробную информацию об этих командах можно, как всегда, получить из Справки Maple.

Сложные типы данных

Сложные типы данных образуются из специальных синтаксических конструкций из выражений Maple (последовательность, список, множество) или обращением к специальной функции-конструктору (массив, таблица).

Последовательность, список и множество

Последовательность выражений — это группа выражений Maple, разделенных запятыми. Например,

```

> s:=x^2, cos(x), ln(x)+y, x^y;
      s := x2, cos(x), ln(x) + y, xy
> whattype(s);
      exprseq
> s[3];
      ln(x) + y

```

Последовательность, таким образом, можно мыслить, как последовательность выражений, перенумерованных натуральными числами. Доступ к значению любого элемента последовательности осуществляется использованием индексной записи — после имени переменной в квадратных скобках указывается индекс (номер) элемента. Однако присвоить новое значение элементу последовательности с использованием индексной формы обращения нельзя.

Для создания длинных последовательностей, элементы которых подчиняются определенной закономерности, можно использовать команду `seq()`. Например,

```

> seq(sin(i*Pi/6), i=0..10);
      0, 1/2, sqrt(3)/2, 1, sqrt(3)/2, 1/2, 0, -1/2, -sqrt(3)/2, -1, -sqrt(3)/2

```

Можно также использовать и операцию повторения `$`

```

> a[i] $ i=1..10;
      a1, a2, a3, a4, a5, a6, a7, a8, a9, a10

```

Список — упорядоченная последовательность выражений, заключенных в квадратные скобки.

Множество — неупорядоченная последовательность выражений, заключенных в фигурные скобки. Этот объект в Maple понимается точно так же, как множество в математике. Если в последовательности присутствуют повторяющиеся элементы, то в множестве им будет соответствовать только один элемент, тогда как в списке все повторяющиеся элементы существенны. Список может содержать повторяющиеся элементы в разных местах.

Списки, как и последовательности, сохраняют порядок своих элементов, поэтому с помощью индекса можно получить значение любого элемента списка, что, впрочем, справедливо и для множества. Более того, можно присвоить новое значение элементу списка с использованием индексной формы обращения, изменив тем самым значение соответствующего элемента списка. Однако присвоить новое значение элементу множества с использованием индексной формы обращения нельзя.

Чтобы изменить элемент множества, его следует удалить операцией `minus`, а затем добавить новый элемент операцией `union`, семантика которых соответствует их аналогам в математике — разности и объединению. Кроме того, в Maple реализована операция пересечения двух множеств `intersect`. Например,

```

> ({a,b,c} minus {b}) union {x};

```

```

                                {x, a, c}
> {a,b,c} intersect {b,c,d};
                                {b, c}

```

При создании множества порядок его элементов формируется системой Maple и может меняться от сеанса к сеансу при работе с одним и тем же рабочим листом.

Массивы и таблицы

Массив представляет собой дальнейшее развитие концепции списка. Если список представляет собой перенумерованную последовательность элементов, индексы которых могут принимать только положительные значения, причем нумерация обязательно начинается с единицы, то в массиве каждый элемент также связан с индексом, однако этот индекс не ограничен одной размерностью. Более того, изменение индекса массива не ограничено положительными целыми значениями, его значение может быть как отрицательным целым числом, так и нулем.

Для объявления массива следует использовать функцию `array()`. Синтаксис этой функции следующий:

```
> array( indexfcn, bounds, list);
```

Параметр `indexfcn` должен быть именем процедуры, задающим правила индексации. Встроенные значения `symmetric`, `antisymmetric`, `sparse`, `diagonal`, `identity` позволяют задать симметричную, кососимметричную, разреженную, диагональную и единичную матрицы. Параметр `bounds` представляет диапазон(ы) изменения индекса(ов) массива. Если массив многомерный, то соответствующие диапазоны должны задаваться подряд через запятую. Значения элементов массива задаются параметром `list`, причем для двумерного массива элементами списка являются списки, содержащие значения соответствующих строк массива. Таким образом, параметр `list` является списком списков. Для массивов большей размерности он представляет собой вложенные списки с глубиной, соответствующей количеству размерностей.

Значения элементов массива можно задать с использованием индексной формы, причем в квадратных скобках следует задать список индексов, соответствующих элементу, которому присваивается значение.

Для отображения в области вывода значений элементов массива можно использовать команду `print()`. Например,

```
> A := array(1..5,1..5,identity):print(A);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> B:=array(1..3,1..3);B[2,3]:=sin(x);
```

```
B := array(1..3,1..3,[ ])
```

$$B_{2,3} := \sin(x)$$

```
> print(B);
```

$$\begin{bmatrix} B_{1,1} & B_{1,2} & B_{1,3} \\ B_{2,1} & B_{2,2} & \sin(x) \\ B_{3,1} & B_{3,2} & B_{3,3} \end{bmatrix}$$

Таблица — дальнейшее развитие массива. В ней в качестве индекса можно использовать не только целые числа, а любые имена. Таблица является удобным объектом, когда надо в одном «массиве» хранить данные, относящиеся к какому-либо объекту и ссылаться к ним по индексам, представляющим естественную запись их наименований. Например, таблицу, содержащую физико-механические характеристики стали, можно задать как

```
> steel:=table([
  density=[7.85*10^3, kg/m^3],
  Young=[2.1*10^5, MPa],
  Poisson=[0.3]]);
```

```
steel :=
```

```
table([Poisson = [0.3], density = [7850.00,  $\frac{kg}{m^3}$ ], Young = [210000.0, MPa ]])
```

```
> steel[density];
```

$$\left[7850.00, \frac{kg}{m^3} \right]$$

Вычисление выражений и переменных сложных типов

Maple по умолчанию стремится производить все преобразования и вычисления с максимальной точностью, то есть там, где возможно, использовать арифметику с рациональными числами. Иногда это не совсем удобно, и необходимо получить результат в форме привычных десятичных дробей. Для вычисления и преобразования коэффициентов выражений, представленных обыкновенными дробями, существует команда `evalf()`, аппроксимирующая дробные числа десятичными числами, а также вычисляющая выражения и константы Maple в форме числа с плавающей точкой. Например,

```
> evalf(Pi);
```

```
3.141592654
```

По умолчанию используется арифметика чисел с плавающей точкой с 10 значащими цифрами. Это значение можно изменить, переустановив системную константу `Digits` в другое значение, например,

```
> Digits:=15;
```

```
Digits := 15
```

```
> evalf(Pi);
```

3.14159265358979

После выполнения этого оператора все последующие в сеансе Maple вычисления с плавающей точкой будут производиться с 15 значащими цифрами. Обратим внимание, что имя этой системной переменной начинается с прописной буквы.

Синтаксис команды `evalf()` следующий:

```
evalf(f);
```

```
evalf(f, n);
```

```
evalf[n](f);
```

где f — вычисляемое выражение, а n — число значащих цифр, используемых при вычислении. Например,

```
> evalf(Pi, 3);
```

3.14

```
> 5/3*exp(-2+3*I)*sin(Pi/4);evalf[15](%);
```

$$\frac{5}{6} e^{(-2+3I)} \sqrt{2}$$

-0.157898022493763 + 0.0225078172647505 I

Maple вычисляет значение переменной простого типа равным тому алгебраическому выражению, которое ей присвоено, но значения переменных сложных, или составных типов (массивы, таблицы, векторы), вычисляются в Maple равным имени этой переменной. Для того, чтобы выяснить действительное значение, содержащееся в такой переменной, в Maple используется команда `eval()`.

Например, переменная, содержащая массив, является переменной сложного типа, и для отображения ее содержимого можно, наряду с командой `print()` использовать команду `eval()`. Например,

```
> a:=array(1..3,1..3,[[1,2,3],[4,5,6],[7,8,9]]);
```

$$a := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

```
> a;
```

a

```
> eval(a);
```

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

Как видно из приведенного примера, простое задание в области ввода имени переменной, содержащей массив, приводит к отображению в области вывода имени самой переменной.

Данную команду можно использовать с переменными любого сложного типа для выяснения их содержимого.

Команда `eval()` может использоваться как команда подстановки, например,

```
> poly := x^3+3*x+2;
                                poly := x3 + 3 x + 2
> eval(poly, x=1);
                                6
> poly := x*y+3*z+2;
                                poly := x y + 3 z + 2
> eval(poly, [x=2, y=3, z=t]);
                                8 + 3 t
> expr := sin(x)/cos(x);
                                expr :=  $\frac{\sin(x)}{\cos(x)}$ 
> subs(x=0, expr);
                                 $\frac{\sin(0)}{\cos(0)}$ 
> eval(expr, x=0);
                                0
```

Обратите внимание на разницу в выполнении команд `eval()` и `subs()` в последнем примере.

Решение уравнений, неравенств и их систем

Практически ни одна математическая задача не обходится без решения какого-нибудь уравнения, системы уравнений, неравенства или системы неравенств. В Maple имеется несколько команд для решения таких задач. Здесь мы познакомимся только с двумя такими командами. Но сначала кратко опишем еще два типа данных, используемых в Maple.

Два выражения, соединенные знаком равенства `=`, представляют новый тип данных Maple — *уравнение* (`equation`). Уравнения можно присваивать обычным переменным Maple, с ними можно осуществлять преобразования, используя обычные арифметические действия, которые выполняются отдельно для левой и правой частей уравнения.

При проверке типа переменной, значением которой является уравнение, с помощью команды `whattype()` результатом является равенство `=`, означающее, что тип проверяемой переменной является уравнением.

Точно также два выражения, соединенные знаками `>=` (больше или равно), `<=` (меньше или равно), `>` (больше) или `<` (меньше), представляют новый тип — *неравенство* (`inequation`).

При проверке с помощью команды `whattype()` типа объекта, представляющего неравенство, результатом будет либо `<>`, либо `<`, либо `<=`.

Команда `solve()`

Команда `solve()` одна из самых полезных команд системы аналитических вычислений Maple. Эта универсальная команда позволяет решать уравнения, системы уравнений, неравенства и системы неравенств. Ее синтаксис, как и синтаксис всех команд Maple, достаточно прост и легко запоминается:

```
solve(eqn, var);  
solve({eqn1,eqn2,...},{var1, var2,...});
```

Первая форма команды предназначена для решения одного уравнения `eqn` относительно заданной переменной `var`, тогда как вторая форма позволяет решать системы уравнений относительно заданных во втором параметре переменных. Обратим внимание на то, что система уравнений и ее неизвестные переменные задаются в виде множеств. Результатом в этом случае является также множество значений неизвестных в виде уравнений, тогда как в случае задания одного уравнения результатом будет выражение (в случае одного корня уравнения) или последовательность выражений (в случае нескольких корней). Если не задана переменная (переменные), относительно которых следует решать уравнение (систему уравнений), то Maple выдаст все решения относительно всех неопределенных переменных в исходных уравнениях.

В общем случае полиномиальное уравнение степени выше 4 может не иметь решения, выраженного с помощью радикалов. В этом случае для представления результатов Maple использует специальную функцию `RootOf(expr, var)`, которая применяется для обозначения любого корня `expr` относительно заданной `var`.

Если решение уравнения выражается достаточно сложной формулой, то Maple может ввести специальные «промежуточные» переменные вида `%n`, где `n` является целым числом, для обозначения каких-либо выражений и использовать эти переменные в окончательном представлении решения. Эти переменные можно в дальнейшем использовать, как обычные переменные с присвоенными значениями, как если бы пользователь задал их сам. Например,

```
> eqn:=x^5+3*x^4-x^3+2*x^2-x+1=0;  
      eqn := x5 + 3 x4 - x3 + 2 x2 - x + 1 = 0  
> sols:=solve(eqn, x);
```

```
sols := RootOf(%1, index = 1), RootOf(%1, index = 2), RootOf(%1, index = 3),
      RootOf(%1, index = 4), RootOf(%1, index = 5)
```

```
%1 := _Z5 + 3 _Z4 - _Z3 + 2 _Z2 - _Z + 1
```

```
> evalf(sols[1]); evalf(sols[2]);
  evalf(sols[3]); evalf(sols[4]);
  evalf(sols[5]);
```

```
0.4602207483 + 0.5002964274 I
```

```
-0.2189557445 + 0.7572686885 I
```

```
-3.482530008
```

```
-0.2189557445 - 0.7572686885 I
```

```
0.4602207483 - 0.5002964274 I
```

В этом примере функция `RootOf()` представляет любое решение уравнения $_Z^5 + 3_Z^4 - _Z^3 + 2_Z^2 - _Z + 1 = 0$. Обратим внимание на переменную `_Z`. Это системная переменная, сгенерированная Maple, которая всего лишь заменяет переменную `x` исходного уравнения. Опция `index` со значением, равным целому числу, служит для нумерации и упорядочивания корней уравнения. С помощью функции `evalf()` можно получить все приближенные корни уравнения.

Для получения всех решений тригонометрических уравнений следует задать значение глобальной переменной `_EnvAllSolutions` равным `true`. По умолчанию Maple решает тригонометрическое уравнение на промежутке $[-\pi, \pi]$. Следующий пример иллюстрирует использование глобальной переменной `_EnvAllSolutions`.

```
> eqn:=cos(x)^3-sin(x)^2+1=0;
```

```
eqn := cos(x)3 - sin(x)2 + 1 = 0
```

```
> sols:=solve(eqn, x);
```

```
sols :=  $\frac{\pi}{2}, -\frac{\pi}{2}, \pi$ 
```

```
> _EnvAllSolutions:=true;sols:=solve(eqn, x);
```

```
_EnvAllSolutions := true
```

```
sols :=  $\frac{1}{2}\pi + 2\pi\_Z1\sim, -\frac{1}{2}\pi + 2\pi\_Z2\sim, \pi + 2\pi\_Z3\sim$ 
```

Как видно, в случае `_EnvAllSolutions:=true` Maple действительно выдает все решения тригонометрического уравнения с использованием целочисленных системных переменных `_Z1~`, `_Z2~`, `_Z3~`, в которых знак тильда (~) означает, что на значения этих переменных наложены некоторые ограничения. В данном случае эти переменные могут принимать только целочисленные значения.

При решении систем уравнений, когда ответ получается в виде множества уравнений, в которых левая часть является неизвестной переменной, для того, чтобы присвоить найденные значения переменным, относительно которых решалась система, следует применять команду `assign()`. Эта команда присваивает переменным, стоящим в левой части уравнений из множества решений, значения, равные правым частям. Например,

```
> solve( {x^2*y^2=0, x^2-y^2=1} );
      {x = 1, y = 0}, {x = -1, y = 0}, {x = 0, y = RootOf( _Z^2 + 1, label = _L1)}
> sols:=solve( {x^2*y^2=0, x^2-y^2=1, x<>0} );
      sols := {x = 1, y = 0}, {x = -1, y = 0}
> assign(sols[1]);x;y;
      1
      0
> x:='x';y:='y';assign(sols[2]);x;y;
      x := x
      y := y
      -1
      0
```

Команда `fsolve()`

По умолчанию Maple пытается найти аналитическое выражение для корней уравнения. Если это ему не удастся, то он просто ничего не выводит в области вывода. В подобных случаях (если корни действительно существуют) можно воспользоваться командой `fsolve()`, которая находит численное решение уравнения или системы уравнений. Формат команды отличается от формата команды `solve()` наличием третьего параметра `options`:

```
fsolve(eqns, vars, options);
```

где задание первых двух параметров соответствует заданию аналогичных параметров в команде `solve()`, а параметр `options` может принимать значения, приведенные в таблице 8.

Таблица 8. Значения параметра `options`

	Значение	Описание
<code>complex</code>		Разыскиваются комплексные корни (только для полиномов)
<code>fulldigits</code>		Используется арифметика с максимальной мантиссой
<code>maxsols</code>		Разыскивается n решений (только для

<p>a..b или x=a..b или {x=a..b, y=c..d, ...}</p> <p>avoid=s</p> <p>var=value</p>	<p>полиномов)</p> <p>Задан интервал $[a, b]$, на котором разыскивается решение (во второй форме задания этой опции x обозначает имя неизвестной переменной в уравнении; третья форма используется для решения системы уравнений)</p> <p>Параметр s есть множество уравнений вида $var=value$; этот параметр указывает определенные значения переменных, которые следует исключить при нахождении корней.</p> <p>Устанавливает начальное значение</p> <p>По умолчанию для произвольного уравнения эта функция находит одно решение, но для полиномов определяются все действительные корни. Чтобы найти <i>все</i> корни полинома, включая комплексные, следует задать опцию <code>complex</code>. Например,</p>
--	---

```
> eq:=x^5+2*x^3-x+1=0;
      eq := x5 + 2 x3 - x + 1 = 0
> fsolve (eq, x);
      -0.8783886413
> fsolve (eq, x, complex);
-0.8783886413 , -0.07074661214 - 1.564121432 I, -0.07074661214 + 1.564121432 I,
  0.5099409328 - 0.4520534110 I, 0.5099409328 + 0.4520534110 I
```

Еще примеры:

```
> fsolve (sin (x) , x);
      0.
> fsolve (sin (x) , x=3.1);
      3.141592654
> fsolve (sin (x) , x, avoid={x=0, x=Pi, x=-Pi}, -10..10);
      -6.283185307
```

Во втором примере определяется корень уравнения вблизи точки $x = 3,1$ (начальное приближение к корню); в третьем — определяется корень на интервале $x \in [-10, 10]$ за исключением известных корней $x = 0$, $x = \pi$ и $x = -\pi$.

Дифференцирование и интегрирование

Maple позволяет вычислять обыкновенные и частные производные аналитического выражения по одной или нескольким переменным. Для этой процедуры

предназначены команды `diff()` и `Diff()`. Вторая команда является так называемой *отложенной* (пассивной) командой (`inert command`), которая не вычисляет производную от выражения, а просто отображает математическую запись операции дифференцирования. Результат действия отложенной команды можно присвоить переменной Maple, а в дальнейшем при помощи команды `value()` вычислить его значение. Отложенная форма команды дифференцирования удобна, когда необходимо видеть, какие операции были сделаны для получения необходимого выражения. Как уже отмечалось ранее, кроме отложенной команды дифференцирования в Maple есть и еще ряд отложенных форм некоторых команд, полную информацию о которых можно получить в Справке.

Синтаксис команды дифференцирования следующий:

```
diff(expr, var_1, var_2, ... , var_n);
diff(expr, [var_1, var_2, ... , var_n]);
```

В результате выполнения любой из приведенных команд будет вычислена частная производная n -го порядка от заданного первым параметром выражения по заданным n переменным.

При формировании производных высокого порядка полезен оператор последовательности `$`, который позволяет проще и нагляднее задать производную. Например, для вычисления третьей производной функции $f(x)$ по переменной x можно использовать команду `diff(f(x), x, x, x)`, в которой три раза указано дифференцирование по переменной x , или применить в команде дифференцирования оператор последовательности `x$3`, что упрощает и делает более наглядным задание третьей производной: `diff(f(x), x$3)`. Несколько примеров использования команды дифференцирования приведены ниже:

```
> f:=x^3+sqrt(ln(x^2+1))+sqrt(y)*ln(cos(x));
```

$$f := x^3 + \sqrt{\ln(x^2 + 1)} + \sqrt{y} \ln(\cos(x))$$

```
> diff(f, x);
```

$$3x^2 + \frac{x}{\sqrt{\ln(x^2 + 1)}(x^2 + 1)} - \frac{\sqrt{y} \sin(x)}{\cos(x)}$$

```
> diff(f, x$2);
```

$$6x - \frac{x^2}{\ln(x^2 + 1)^{(3/2)}(x^2 + 1)^2} + \frac{1}{\sqrt{\ln(x^2 + 1)}(x^2 + 1)} - \frac{2x^2}{\sqrt{\ln(x^2 + 1)}(x^2 + 1)^2} - \sqrt{y} - \frac{\sqrt{y} \sin(x)^2}{\cos(x)^2}$$

```
> diff(f, x, y);
```

$$\frac{1}{2} \frac{\sin(x)}{\sqrt{y} \cos(x)}$$

```
> Diff(f, x, y);
```

$$\frac{\partial^2}{\partial y \partial x} (x^3 + \sqrt{\ln(x^2 + 1)} + \sqrt{y} \ln(\cos(x)))$$

> Diff(f, y\$2); value(%);

$$\frac{\partial^2}{\partial y^2} (x^3 + \sqrt{\ln(x^2 + 1)} + \sqrt{y} \ln(\cos(x)))$$

$$\frac{1}{4} \frac{\ln(\cos(x))}{y^{(3/2)}}$$

Интегрирование выражений по заданной переменной осуществляется командой `int()`, которая также имеет отложенную форму `Int()`. Эта команда позволяет вычислять как неопределенный интеграл от выражения (при этом в ответе не будет никакой постоянной интегрирования) с использованием следующего синтаксиса команды:

`int(expr, var);`

так и определенный интеграл с помощью следующего синтаксиса команды

`int(expr, var=a..b);`

где `a` и `b` являются пределами интегрирования, причем эти пределы могут быть и аналитическими выражениями. Несколько примеров использования команды интегрирования приведены ниже:

> f:=a*x^3*cos(b*x);

$$f := a x^3 \cos(b x)$$

> int(f, x);

$$\frac{a (b^3 x^3 \sin(b x) + 3 b^2 x^2 \cos(b x) - 6 \cos(b x) - 6 b x \sin(b x))}{b^4}$$

> Int(f, x=0..Pi);

$$\int_0^{\pi} a x^3 \cos(b x) dx$$

> value(%);

$$\frac{a (6 - 6 b \sin(\pi b) \pi - 6 \cos(\pi b) + b^3 \sin(\pi b) \pi^3 + 3 b^2 \cos(\pi b) \pi^2)}{b^4}$$

Для символьного вычисления определенного интеграла существуют две опции, управляющие обработкой разрывов подынтегральной функции. Эти опции задаются третьим параметром в командах `int()` или `Int()`.

По умолчанию команда интегрирования проверяет выражение на непрерывность в области интегрирования и вычисляет интеграл как сумму отдельных определенных интегралов на промежутках непрерывности функции. Опция ``continuous`` отключает этот режим и вычисляет интеграл как разность значений первообразной подынтегральной функции в точке начала и конца промежутка интегрирования.

Еще одна опция `CauchyPrincipalValue` вычисляет несобственные интегралы первого и второго рода в смысле главного значения Коши.

Если Maple не может найти замкнутую форму выражения для определенного интеграла, то команда интегрирования возвращает просто вызов самой себя (в области вывода печатается математическая запись вычисления интеграла, как при обращении к отложенной команде интегрирования). В подобных случаях можно вычислить значение определенного интеграла численным способом с помощью команды `evalf()`. Синтаксис подобной конструкции может быть следующий:

```
evalf( Int(f, x=a..b) );
```

При решении задачи численного интегрирования можно определить дополнительные опции в одной из следующих форм: `method = <name> (или <name>); digits = <posint> (или <posint>); epsilon = <numeric>`.

Опция `method = <name> (или просто <name>)` указывает специфический метод численного интегрирования, который нужно применить. По умолчанию, как отмечают разработчики системы, используется гибридная символьно-численная стратегия.

Опция `digits = <posint> (или просто <posint>)` указывает число цифр точности для вычисления. По умолчанию точность для вычислений определяет переменная среды Maple `Digits`.

Опция `epsilon = <numeric>` определяет допуск относительной погрешности для вычисленного результата.

Несколько примеров численного вычисления интегралов помогут освоиться с этой методикой:

```
> int(sin(x)*ln(x), x=0..1);
```

$$-\gamma + \text{Ci}(1)$$

```
> Int(sin(x)*ln(x), x=0..1)=evalf(Int(sin(x)*ln(x), x=0..1));
```

$$\int_0^1 \sin(x) \ln(x) dx = -0.2398117420$$

```
> Int(exp(x-x^2/2)/(1+exp(x)/2), x=-infinity..infinity)=  
evalf(
```

```
Int(exp(x-x^2/2)/(1+exp(x)/2), x=-infinity..infinity));
```

$$\int_{-\infty}^{\infty} \frac{e^{(x-1/2)x^2}}{1 + \frac{1}{2}e^x} dx = 1.805577062$$

Более подробную информацию можно найти в Справке Maple.

С другими командами, например, командами, решающими дифференциальные уравнения, как обыкновенные, так и в частных производных, или командами,

выполняющими интегральные преобразования, мы познакомимся на примерах решения конкретных задач математической физики в последующих главах книги.

Пакеты

Пакеты в Maple используются для удобства организации работы пользователя. Пакет представляет собой набор команд для решения задач, относящихся к определенным разделам математики, или решения определенных задач графического представления информации. Например, пакеты `LinearAlgebra` и `VectorCalculus` служат для решения задач линейной алгебры и векторного исчисления, в пакете `Student` собраны команды, необходимые при обучении стандартных разделов математики и т.д.

Для того чтобы использовать команды какого-нибудь пакета, необходимо подключить его, так как они все находятся не в ядре системы Maple, а в специальных файлах. Подключение пакетов осуществляется с помощью команды

```
with(пакет);
```

где в качестве параметра указывается имя соответствующего пакета. Может оказаться, что подключаемый пакет содержит команду с таким же именем, что и в ранее подключенном пакете. В этом случае в области вывода отображается сообщение о переопределении соответствующей команды, и результат ее действия будет соответствовать команде, находящейся в последнем подключенном пакете.

Полный список и описание пакетов можно найти в справочной системе Maple, или выполнить команду

```
?index[package];
```

С некоторыми командами пакета графического вывода результатов `plots` мы познакомимся в разделе *Графика*, с командами других пакетов будем знакомиться по мере необходимости их использования для решения конкретных задач.

Графика

Системы аналитических вычислений привлекают исследователей не только своими возможностями реализации алгоритмов построения аналитических решений, но и развитой графикой, начиная от построения простейших двумерных кривых и заканчивая сложными трехмерными поверхностями и анимацией двумерных и трехмерных изображений. В любой момент пользователь может отобразить результаты своих исследований в виде графических образов, которые, как известно, более информативны, чем скупые ряды цифр. (Хотя в некоторых ситуациях цифра может оказаться более полезной, чем общая картина изменения какого-либо параметра.)

Исходными данными для графических команд, объединенных в пакете `plots`, могут выступать как типы данных, поддерживаемые Maple (функции, массивы, множества, графические структуры), так и данные других программ и приложений, сохраненные в файлах текстового формата.

Чтобы можно было использовать команды графического отображения этого пакета, необходимо подключить его с помощью оператора `with(plots)`.

Простейшая графическая команда `plot()` расположена в системной библиотеке Maple, и поэтому доступна в любое время. Именно с нее мы и начнем знакомство с графическими возможностями системы аналитических вычислений Maple.

Команды двумерной графики

Наиболее употребительной командой двумерной графики является многофункциональная команда `plot()`. Она позволяет рисовать график одной или нескольких функций одной переменной, заданных в явном или параметрическом виде, а также отображать множество точек в декартовой или полярной системе координат. Синтаксис команды `plot()` следующий

```
plot(f, h, v, опции);
```

где `f` — функция, которую необходимо отобразить, `h`, `v` — соответственно диапазон изменения независимой переменной по горизонтальной оси графика и диапазон изменения значения функции по вертикальной оси графика. Отметим, что диапазон изменения независимой переменной задается в виде `x=a..b`, где `a` и `b` наименьшее и наибольшее значение изменения переменной, а `x` имя независимой переменной. Если диапазон не задан, то есть второй параметр представляет собой просто имя независимой переменной в функции, то по умолчанию принимается следующий интервал ее изменения `-10..10`. Этот (второй) параметр обязательно должен присутствовать при задании графика командой `plot()`. Вертикальный диапазон ограничивает вывод графика заданной областью изменения функции. Этот параметр необязателен, как и опции, задающиеся в виде уравнений `имя_опции=значение`. При отсутствии явного задания опций принимаются значения опций по умолчанию.

Опции задают вид отображаемого графика: толщину, цвет и тип линии графика, вид осей координат, размещение надписей и т.д. Набор возможных опций во всех командах двумерного графического вывода, за некоторым исключением, одинаков. Ниже перечислены некоторые из наиболее употребительных опций и соответствующие им значения:

- `title='строка'` — задает заголовок рисунка. По умолчанию заголовок не выводится.
- `coords=<значение>` — по умолчанию при выводе как явно заданной функции, так и параметрически заданной функции используется декартова система координат. Задание значения данной опции меняет тип системы координат. Возможные следующие значения: `bipolar`, `cardiod`, `cassinian`, `elliptic`, `hyperbolic`, `invcassinian`, `invelliptic`, `logarithmic`, `logcosh`, `maxwell`, `parabolic`, `polar`, `rose` и `tangent`, описание которых можно получить в справочной системе Maple с помощью команды `?coords`. Здесь отметим только, что значение `polar` задает полярную систему координат.
- `axes=<значение>` — определяет тип отображаемых осей координат. Эта опция может принимать следующие значения: `normal` — обычные оси координат; `boxed` — график заключен в прямоугольник с нанесенными

шкалами по нижней и вертикальной левой граням; `frame` — оси с точкой пересечения в левом нижнем углу рисунка; `none` — оси не отображаются.

- `scaling=<значение>` — масштаб, в котором отображается график. Если значение этой опции равно `constrained`, то это соответствует заданию абсолютных значений по осям координат, то есть одна единица измерения по оси независимой переменной равна одной единице измерения по оси значений функции. Умалчиваемое значение равно `unconstrained`, и это соответствует тому, что оси растягиваются таким образом, чтобы их размеры соответствовали размерам графического окна вывода.
- `style=<значение>` — задание отображения графика функции линиями (значение опции равно `line`) или точками (значение опции равно `point`). Значения параметра `patch` и `patchnogrid` применяются только тогда, когда выводится замкнутый многоугольник. В этом случае его внутренняя область закрашивается цветом, установленным в опции `color`. Если в графическом выводе нет замкнутых многоугольников, то действие этих значений опции соответствуют значению `line`.
- `numpoints=n` — число вычисляемых точек, по которым строится график (значение по умолчанию равно 50)
- `resolution=n` — горизонтальное разрешение дисплея в пикселях (значение по умолчанию равно 200)

- `color=<значение>` — задание цвета, которым отображается график. В качестве значения этой опции может выступать одно из зарезервированных значений цвета в Maple: `aquamarine`, `black`, `blue`, `navy`, `coral`, `cyan`, `brown`, `gold`, `green`, `gray`, `grey`, `khaki`, `magenta`, `maroon`, `orange`, `pink`, `plum`, `red`, `sienna`, `tan`, `turquoise`, `violet`, `wheat`, `white` и `yellow`. Можно также определить и свой собственный цвет, соответствующий смещению заданных частей красного, зеленого и синего цветов. Это осуществляется с помощью следующей команды

```
macro (palegreen=COLOR (RGB, .5607, .7372, .5607))
```

где `palegreen` имя константы нового цвета, в котором красный составляет 0.5607 части, зеленый — 0.7372 и синий — 0.5607.

- `font=<значение>` — задание шрифта для вывода текста на рисунке. Значение опции задается в виде списка [семейство, стиль, размер]. Параметр `семейство` задает гарнитуру шрифта: `TIMES`, `COURIER`, `HELVETICA` или `SYMBOL`. Параметр `стиль` определяет стиль шрифта: для гарнитуры `TIMES` возможные значения `ROMAN`, `BOLD`, `ITALIC` или `BOLDITALIC`, для гарнитур `COURIER` и `HELVETICA` стиль можно опустить, или задать `BOLD`, `OBLIQUE` или `BOLDOBLIQUE`, для шрифта `SYMBOL` стиль не задается. Последний параметр `размер` задает размер шрифта в пунктах (один пункт приблизительно равен 1/72 дюйма).

- `axesfont=<значение>` — задание шрифта для надписей под засечками вдоль осей координат. Значение этой опции аналогично значению опции `font`.
- `labelfont=<значение>` — задание шрифта для названия осей координат. Значение этой опции аналогично значению опции `font`.
- `titlefont=<значение>` — задание шрифта для заголовка рисунка. Значение этой опции аналогично значению опции `font`.
- `linestyle=n` — задание типа линии графика. Значением этой опции является целое число `n`. При `n=0` или `1` график отображается сплошной линией. Если значение больше `2`, то используются разнообразные шаблоны линий для вывода графика функции: штриховые, с разной толщиной и т.д.
- `thickness=n` — задание толщины линии графика, меняется от `0` до `15`, соответствуя изменению толщины линии от самой тонкой до самой жирной.
- `symbol=<значение>` — задание типа символа, которым помечаются точки графика функции при `style=point`. Может принимать следующие значения: `box`, `cross`, `circle`, `POINT` (прописными буквами) и `diamond`.
- `labels=[x,y]` — задание названий осей координат. Параметры `x` и `y` задаются в виде строк и соответствуют отображаемым названиям горизонтальной и вертикальной осей. По умолчанию принимают значения имени независимой переменной и имени функции.
- `xtickmarks=n` — задает число точек, не менее которых должно быть помечено на горизонтальной оси. Значение этой опции может быть целым числом или списком значений координат точек горизонтальной оси.
- `ytickmarks=n` — задает число точек, не менее которых должно быть помечено на вертикальной оси. Значение этой опции может быть целым числом или списком значений координат точек вертикальной оси.

Большинство этих опций можно изменить с помощью команд меню интерфейса пользователя или команд контекстного меню, отображаемого нажатием правой кнопки мыши в области рисунка. Как отмечалось ранее, при выделении в тексте документа графики или получении фокуса окна с графическим выводом, если задан режим вывода графики в отдельном окне, меняется основное меню в строке меню интерфейса, а также заменяется контекстная панель инструментов таким образом, чтобы обеспечить доступ к командам интерфейса, работающим с полученным графическим отображением.

Кратко остановимся на описании интерфейса пользователя с меню и контекстной панелью инструментов для работы с графикой.

При выделении графики в строке меню появляются новые меню **Style** (Стиль), **Legend** (Надпись), **Axes** (Оси), **Projection** (Проекция), **Animation** (Анимация) и **Export** (Экспорт), которые позволяют изменить основные опции для вывода графики. Все команды этих меню дублируются в контекстном меню, а часть в контекстной панели инструментов для двумерной графики.

Команды **Line** (Линия), **Point** (Точка), **Patch** (Заливка) и **Patch w/o grid** (Заливка без сетки) меню **Style** устанавливают значение опции `style` равной соответственно `line`, `point`, `patch` или `patchnograd`.

Подменю **Symbol** (Символ), **Line Style** (Тип линии) и **Line Width** (Толщина линии) позволяют установить значения опций `symbol`, `linestyle` и `thickness`. Кнопки на контекстной панели инструментов для этих опций не предусмотрено.

Команды меню **Axes** соответствуют установке значений `boxed`, `frame`, `normal` и `none` опции `axes`.

Меню **Projection** устанавливает значения опции `scaling`. На панели инструментов значения этой опции можно устанавливать с помощью соответствующей кнопки. Если она не нажата, то соответствует значению `unconstrained`, если нажата — установлено значение `constrained`.


Меню **Animation** специально предназначено для анимации изображений и становится доступным, когда в документе Maple графика выводится командой создания анимации `animate()`.

Меню **Export** предназначено для экспорта графики в файлы различных форматов.

Все перечисленные команды меню можно выполнить из контекстного меню (открывающегося при нажатии правой кнопки мышки в области рисунка), в котором кроме этих команд присутствует команда **Copy** (Копировать), копирующая графический рисунок в буфер обмена операционной системы Windows для вставки его в документ другого приложения или обработки какой-либо графической программой.


Несколько примеров помогут освоиться с этой простой в использовании командой отображения графики:

```
> plot(x^2+cos(x^2), x=-3..3,
      color=black,
      title=`График функции`,
      titlefont=[HELVETICA, 16],
      xtickmarks=8,
      thickness=2,
      axesfont=[COURIER, 11],
      labels=[`x`, `Y`],
      labelfont=[COURIER, BOLD, 11]);
```



Для вывода нескольких функций на одном графике необходимо в команде `plot()` задавать функции в виде множества, а значение опции `color` в виде списка позволяет задать цвет для вывода графиков функций:


```
> plot({x^2+sin(x^3), 20*exp(-2*x)*sin(x)},
       x=-1..3.5, -1..5.5,
       color=[black,red],
       title=`Вывод нескольких графиков`,
       titlefont=[HELVETICA,16],
       xtickmarks=8,
       thickness=2,
       axesfont=[COURIER,11],
       labels=[`Аргумент`, `Функции`],
       labelfont=[COURIER,BOLD,11]);
```



Две функции $y = x^2 + \sin(x^3)$ и $y = 20 e^{-2x} \sin x$ заданы в виде множества. Значение опции `color` определяет цвет графика первой функции черным, а второй — красным.

Параметрическая кривая задается в виде списка, где первый и второй элементы являются значениями соответственно горизонтальной и вертикальной координат, а третий элемент списка задает изменение параметра в принятом в Maple виде. Следующий пример демонстрирует отображение параметрически заданной кривой:

```
> plot([cos(t)^3, sin(t)^3, t=0..2*Pi],
       color=[black],
       title=`Астроида`,
       titlefont=[HELVETICA, 12],
       xtickmarks=10,
       ytickmarks=10,
       thickness=2,
       axesfont=[COURIER, 11]);
```




Команда `plot()` позволяет отображать на графике отдельные точки, которые задаются в виде списка списков, то есть списка, элементами которого являются списки. Эти элементы задают координаты точек по горизонтали и по вертикали. Для вывода точек необходимо задать значение опции `style` равной `POINT`. Если этого не сделать, то Maple отобразит ломаную линию, соединяющую точки в последовательности их задания. Следующий пример отображает точки на плоскости в виде символа окружность `symbol=circle`:

```
> plot([[1.5, 0.5], [2.5, 1], [3, 4], [-2, 3], [-1, 4], [0.5, 3], [1, 2]],
       x=-3..3, 0..5,
       color=[black],
       style=POINT,
       symbol=circle,
       title=`Вывод точек командой plot`,
       titlefont=[HELVETICA, 16],
```

```

xtickmarks=4,
symbolsize=25,
axesfont=[COURIER,11],
labels=[`x`,`y`],
labelfont=[COURIER,BOLD,11]);

```



В пакете `plots` существует большой набор команд для отображения кривых на плоскости. Со всеми этими командами можно познакомиться по справочной системе Maple. Отметим некоторые возможности двумерного графического вывода в системе Maple:

- построение графика функции в логарифмическом масштабе по вертикальной оси (`logplot()`),
- построение графика функции в логарифмическом масштабе по обеим осям координат (`loglogplot()`),
- построение замкнутого многоугольника, заданного своими вершинами (`polygonplot()`),
- изображение конформного отображения комплексной функции (`conformal()`),
- вывод текстовой строки (`textplot()`),
- построение графика неявно заданной функции двух переменных (`implicitplot()`),
- изображение решения дифференциального уравнения (`odeplot()`)

и другие. Все эти команды становятся доступными после подключения пакета `plots` командой `with(plots)`.


Часто необходимо на одном графике совместить результаты отображения, осуществляемого разными графическими командами. Для этого результат действия каждой команды должен быть присвоен некоторой переменной. Заметим, что при этом никакого графического вывода не происходит. Команда `display()`, находящаяся в пакете `plots`, выводит «отложенную» и сохраненную в переменных Maple графику. Синтаксис этой команды следующий:

```
display([pic1, pic2, ...], <опции>);
```

Здесь элементы списка являются либо явно заданными графическими образами с помощью команд вывода графики, либо переменные, в которых хранятся графические образы. Опции соответствуют рассмотренным ранее опциям команды `plot()`. Следует отметить опцию `insequence`, которая применима только для этой команды. Если ее значение равно `true`, то графические образы `pic1`, `pic2`, ... будут последовательно отображаться один за другим, составляя таким образом кадры анимации.

Следующий пример дает представление о том, как используется команда `display()`. В переменной `pic1` сохраняется график функции $\sin(x)$, переменная `pic2` содержит вывод выносной линии для надписи, а переменная `pic3` хранит надпись, выводимую рядом с выносной линией.

```
> with(plots):  
pic1:=plot(sin(x), x=0..2*Pi, thickness=2):  
pic2:=plot([[Pi/2, 1], [Pi/2+Pi/4, 1.45]], color=black):  
pic3:=textplot([[Pi/2+Pi/4, 1.5, `Точка экстремума`]],  
               color=blue, align=RIGHT,  
               font=[COURIER, 12]):  
display([pic1, pic2, pic3]);
```



Результирующий график, составленный из этих трех графических изображений, выводится с помощью команды `display()`, в которой в качестве элементов списка заданы вышеупомянутые переменные, хранящие графические образы.

Элементы программирования в Maple

Программирование занимает особое место в Maple. Более чем 90% встроенных команд Maple запрограммированы на собственном языке программирования Maple, подобном языкам высокого уровня, таким как, скажем старый и добрый Фортран. Язык программирования Maple позволяет создавать программы как символьных, так и численных расчетов любой сложности. Однако для первоначального знакомства с системой Maple язык программирования Maple изучать совершенно не обязательно.

Возможности Maple позволяют решить подавляющее большинство задач без всякого программирования, да к тому же, как правило, несколькими способами.

И, тем не менее, при решении некоторых задач полезно бывает использовать те или иные возможности языка программирования Maple. В последующих главах книги есть примеры решения задач с элементами программирования. Здесь мы рассмотрим самые элементарные средства программирования — создание собственных функций, процедур, условный оператор и операторы цикла. Этого оказывается вполне достаточно, чтобы эффективно решать довольно сложные задачи математической физики.

Функции и процедуры

Функции (на языке программирования — процедуры-функции или операторы-функции) в Maple можно задавать несколькими способами. Один из них — с помощью функционального оператора — специальной формы процедуры. Функциональный оператор задается с помощью «стрелки» \rightarrow : `vars` \rightarrow `result`. Здесь `vars` последовательность имен переменных или одна переменная, `result` — порядок (процедура) действий над переменными `vars`. Например,

```
> f := (x, y) -> exp(-x^2) * cos(y);
```

$$f := (x, y) \rightarrow e^{(-x^2)} \cos(y)$$

```
> f(1.1, 0);
```

0.2981972794

```
> f(2/3, 3);
```

$$e^{(-4/9)} \cos(3)$$

Другой способ создания функционального оператора — использование команды `unapply()`. Простейшая форма этой команды `unapply(expr, x, y, ...)`. Здесь `expr` — выражение, зависящее от переменных `x, y, ...`. Результатом выполнения этой команды будет создание функции от переменных `x, y, ...` из выражения `expr`. Несколько примеров демонстрируют работу этой команды:

```
> p := x^2 + sin(x) + 1;
```

$$p := x^2 + \sin(x) + 1$$

```
> f := unapply(p, x);
```

$$f := x \rightarrow x^2 + \sin(x) + 1$$

```
> f(Pi/6);
```

$$\frac{1}{36} \pi^2 + \frac{3}{2}$$

```
> q := x^2 + y^3 + 1;
```

$$q := x^2 + y^3 + 1$$

```
> f := unapply(q, x);
```

```


$$f := x \rightarrow x^2 + y^3 + 1$$

> f(2);

$$5 + y^3$$

> g := unapply(q, x, y);

$$g := (x, y) \rightarrow x^2 + y^3 + 1$$

> g(2, 3);
32
> h := unapply(q, [x, y]);

$$h := (x, y) \rightarrow x^2 + y^3 + 1$$

> h(2, 3);
32

```

Для создания процедуры (или подпрограммы), которую можно впоследствии использовать неоднократно, в Maple применяется следующая конструкция:

```

имя_процедуры:=proc([список_формальных_параметров])
    [local список_локальных_переменных;]
    [global список_глобальных_переменных]
    [options список_опций;]
    [descriptions строка_описания;]
    последовательность_операторов
end proc;

```

Здесь квадратные скобки в синтаксисе операторов Maple используются для определения не обязательных элементов конструкции. Всякая процедура в Maple начинается с заголовка, состоящего из имени процедуры, за которым следует знак присваивания := и служебное слово proc, затем в круглых скобках через запятую указываются формальные параметры. Процедура обязательно должна завершаться последовательностью ключевых слов end proc (для совместимости с предыдущими версиями в Maple завершать объявление процедуры можно просто ключевым словом end), после которой ставится знак (;) или (:).

При объявлении процедуры единственным обязательным оператором является последовательность операторов, формирующих тело процедуры. Остальные параметры, определяющие локальные и глобальные переменные, список формальных параметров, задающие специальные опции режима работы процедуры и строка описания могут полностью отсутствовать.

После того как процедура загружена, ее вызов осуществляется по имени. Возвращаемым значением по умолчанию является значение последнего оператора из тела процедуры. Тип результата работы процедуры зависит от типа возвращаемого значения.

При описании формального параметра процедуры можно явно указывать его тип после двоеточия, следующего за именем параметра. При таком описании Maple автоматически проверяет тип фактического параметра и выдает сообщение об ошибке, если тип фактического параметра не совпадает с типом формального параметра.

После заголовка процедуры может следовать описательная часть процедуры, отделенная от него пробелом. Для определения локальных переменных, используемых только внутри данной процедуры, применяется описатель `local`, за которым через запятую следуют имена локальных переменных. Перечислить используемые глобальные переменные можно при помощи описателя `global`.

После описателей в процедуре может стоять указатель на опции `options`. В качестве опций используются значения: `remember` — делает более эффективным выполнение рекурсивных процедур, последовательность значений `operator`, `arrow` — аналогична заданию оператора-функции, и некоторые другие, которые можно посмотреть в Справке.

Примеры процедур приводятся в последующих главах книги при решении конкретных задач.

Условный оператор

Обычно программа реализует некоторый алгоритм решения задачи. В таком алгоритме может возникнуть необходимость выполнения определенной последовательности операторов в зависимости от истинности или ложности того или иного выражения. В Maple ветвление в программе реализуется оператором `if`. Общая форма этого оператора имеет вид:

```
if булево_выражение then последовательность_операторов
[elif булево_выражение then последовательность_операторов]
[else последовательность_операторов]
end if;
```

Семантика этого оператора состоит в следующем: если истинно булево выражение, стоящее после ключевого слова `if`, то выполняется последовательность операторов, стоящая после ключевого слова `then`, до первого встретившегося `elif`, `else`, `end if`; если его значение `false` или `FAIL`, то проверяется на истинность булево выражение после ключевого слова `elif` если оно задано, и в случае истинности выполняются операторы после ключевого слова `then`. Если ни одно из булевых условий не истинно, то выполняются операторы блока `else` (в случае его задания). Блоков `elif` может быть сколько угодно, тогда как блок `else` всегда только один.

Синтаксис Maple позволяет использовать вложенные конструкции `if`, то есть последовательности операторов в блоках `then` и `else` могут содержать операторы ветвления.

В системе Maple есть еще один полезный вариант использования операции `if`. Операция ``if`` (обратите внимание — ключевое слово `if` в обратных кавычках) предназначена для использования в выражениях. Она имеет следующий вид:

``if` (условие, операнд_1, операнд_2)`. Смысл этой операции таков: сначала вычисляется значение параметра `условие`, который должен быть булевым выражением, и если он равен `true`, то результатом операции будет `операнд_1`, в противном случае `операнд_2`. Например,

```
> x:=4;y:=2;
                                     x := 4
                                     y := 2
> z:='if` (x>y, x, y)+cos(`if` (y>x, y, x));
                                     z := 4 + cos(4)
> evalf(%);
                                     3.346356379
```

Оператор цикла

Для организации повторяющихся вычислений в Maple предусмотрены следующие формы операторов цикла: `for - from` и `for - in`. Первая форма является универсальной и включает в себя операторы цикла, повторяющиеся заданное число раз, и операторы цикла, выполняющиеся, пока заданное булево выражение является истинным. Вторая форма реализует цикл по элементам списка или множества. Общий синтаксис формы `for - from` следующий:

```
[for имя] [from выражение] [by выражение] [to выражение]
[while булево_выражение]
do последовательность_операторов end do;
```

В этой форме задается имя переменной цикла, блоки `from` и `to` определяют соответственно начальное и конечное значение диапазона изменения переменной цикла, а в блоке `by` задается шаг изменения переменной цикла. Этот шаг может быть и отрицательным. Выполнение цикла начинается с присваивания переменной цикла начального значения, после чего проверяется, не превосходит ли оно конечного значения, и в случае положительного ответа выполняются операторы цикла, заданные в блоке `do...end`, переменная цикла увеличивается на значение шага и алгоритм проверки повторяется снова. Если значение переменной цикла превосходит конечное значение, то цикл прекращает свое выполнение. Если при проверке начального значения переменной цикла, оно превосходит конечное значение, то цикл завершает свое выполнение, и операторы цикла не выполняются ни разу.

Если задан блок `while`, то одновременно с проверкой значения переменной цикла проверяется на истинность булево выражение этого блока, и цикл также завершает работу, если его значение оказывается равным `false` или `FAIL`.

Все перечисленные блоки являются необязательными и могут следовать в произвольном порядке за одним исключением: если задается блок `for`, то он должен следовать первым.

Если какой-то блок не задан, то по умолчанию параметр его принимает значение из таблицы 9:

Таблица 9. Значения параметров по умолчанию

	Блок	Значение
for		Фиктивная переменная
from		1
by		1
to		infinity
while		true

Например, вычисление суммы и произведения можно организовать в цикле

```
> s:=0;p:=1;
```

$$s := 0$$

$$p := 1$$

```
> for i from -3 to 3
```

```
do s:=s+C[i]*x^i; p:=p*(x-a[i]) end do;
```

$$s := \frac{C_{-3}}{x^3}$$

$$p := x - a_{-3}$$

$$s := \frac{C_{-3}}{x^3} + \frac{C_{-2}}{x^2}$$

$$p := (x - a_{-3})(x - a_{-2})$$

$$s := \frac{C_{-3}}{x^3} + \frac{C_{-2}}{x^2} + \frac{C_{-1}}{x}$$

$$p := (x - a_{-3})(x - a_{-2})(x - a_{-1})$$

$$s := \frac{C_{-3}}{x^3} + \frac{C_{-2}}{x^2} + \frac{C_{-1}}{x} + C_0$$

$$p := (x - a_{-3})(x - a_{-2})(x - a_{-1})(x - a_0)$$

$$s := \frac{C_{-3}}{x^3} + \frac{C_{-2}}{x^2} + \frac{C_{-1}}{x} + C_0 + C_1 x$$

$$p := (x - a_{-3})(x - a_{-2})(x - a_{-1})(x - a_0)(x - a_1)$$

$$s := \frac{C_{-3}}{x^3} + \frac{C_{-2}}{x^2} + \frac{C_{-1}}{x} + C_0 + C_1 x + C_2 x^2$$

$$p := (x - a_{-3})(x - a_{-2})(x - a_{-1})(x - a_0)(x - a_1)(x - a_2)$$

$$s := \frac{C_{-3}}{x^3} + \frac{C_{-2}}{x^2} + \frac{C_{-1}}{x} + C_0 + C_1 x + C_2 x^2 + C_3 x^3$$

$$p := (x - a_{-3})(x - a_{-2})(x - a_{-1})(x - a_0)(x - a_1)(x - a_2)(x - a_3)$$

В отличие от других языков программирования, где для организации циклов используется подобная либо оператору `for`, либо оператору `while` конструкция, Maple предлагает программисту ряд команд, в которых реализованы часто используемые в работе циклы. Например, для вычисления суммы и произведения в Maple нет необходимости программировать цикл, а достаточно использовать соответствующие команды

```
> sum(C[k]*x^k, k=-3..3); product(x-a[k], k=-3..3);
```

$$\frac{C_{-3}}{x^3} + \frac{C_{-2}}{x^2} + \frac{C_{-1}}{x} + C_0 + C_1 x + C_2 x^2 + C_3 x^3$$

$$(x - a_{-3})(x - a_{-2})(x - a_{-1})(x - a_0)(x - a_1)(x - a_2)(x - a_3)$$

или

```
> add(C[k]*x^k, k=-3..3); mul(x-a[k], k=-3..3);
```

$$\frac{C_{-3}}{x^3} + \frac{C_{-2}}{x^2} + \frac{C_{-1}}{x} + C_0 + C_1 x + C_2 x^2 + C_3 x^3$$

$$(x - a_{-3})(x - a_{-2})(x - a_{-1})(x - a_0)(x - a_1)(x - a_2)(x - a_3)$$

Отметим еще одну полезную команду `map()`, которая в цикле выполняет вызов функции, определяемой первым параметром команды, и использует в качестве первого параметра функции операнды выражения, определяемого вторым параметром команды `map()`. Например,

```
> map(abs, sum(a[i], i=1..5));
```

$$|a_1| + |a_2| + |a_3| + |a_4| + |a_5|$$

```
> map(diff, [x^n, ln(sqrt(x)), exp(-x^2)], x);
```

$$\left[\frac{x^n n}{x}, \frac{1}{2x}, -2x e^{(-x^2)} \right]$$

Вторая форма оператора цикла организует цикл по элементам объекта, который может быть представлен последовательностью, списком, множеством, суммой, произведением или строкой. Общая форма имеет следующий вид

```
for имя in объект
```

```
[while булево_выражение]
```

```
do последовательность_операторов end do;
```

Переменная цикла, определяемая в блоке `for - in`, последовательно принимает значения операндов объекта, как они определяются командой `op()`. Цикл выполняется столько раз, сколько операндов задано в объекте, если только булево выражение в необязательном блоке `while` не станет ложным раньше, чем будут последовательно исчерпаны все операнды объекта.

Например, суммирование элементов множества можно осуществить с помощью цикла

```
> S:={sin(x), ln(sqrt(x)), exp(2*x)};
      S := {ln(sqrt(x)), sin(x), e(2x)}
> s:=0; for j in S do s:=s+j end do;
      s := 0
      s := ln(sqrt(x))
      s := ln(sqrt(x)) + sin(x)
      s := ln(sqrt(x)) + sin(x) + e(2x)
```

На этом мы завершаем такое очень краткое введение в Maple. В последующих главах при решении различных задач математической физики мы познакомимся с другими возможностями Maple.

2. Обыкновенные дифференциальные уравнения

В приложениях математики к техническим наукам дифференциальные уравнения занимают особо важное место. Они дают возможность решать многие вопросы общетехнических и специальных прикладных дисциплин: физики, теоретической механики, сопротивления материалов, гидравлики, теории машин и механизмов, химии, технологии производств, биологии, финансово-экономических и других дисциплин, — и часто сами возникают при решении этих вопросов. Многочисленные и разнообразные технические приложения теории обыкновенных дифференциальных уравнений требуют в первую очередь знаний различных физико-механических законов.

Для получения как аналитических, так и численных решений обыкновенных дифференциальных уравнений, а также и систем таких уравнений в Maple имеется функция `dsolve`, причем во всех случаях используется единый формат этой функции: `dsolve(ode, var, opts)`. Здесь `ode` — дифференциальное уравнение или система дифференциальных уравнений; `var` — неизвестная функция или функции, если речь идет о системе уравнений; `opts` — не обязательные опции, управляющие решением задачи.

Функция `dsolve` способна успешно решать различные типы задач для обыкновенных дифференциальных уравнений. Это:

- Поиск решений в замкнутой форме обыкновенного дифференциального уравнения или системы таких уравнений;
- Решение обыкновенного дифференциального уравнения или системы таких уравнений с заданными начальными или граничными условиями (задача Коши, краевая задача);
- Поиск решений в виде формального степенного ряд для обыкновенного линейного дифференциального уравнения с полиномиальными коэффициентами;
- Поиск решений с помощью интегральных преобразований (Лапласа, Фурье, и т.п.);
- Поиск численного решения или решения в виде ряда обыкновенного дифференциального уравнения или системы таких уравнений.

В этой главе мы рассмотрим, как с помощью Maple получать общие решения обыкновенных дифференциальных уравнений и систем, как выделять из общего решения частное решение, удовлетворяющее тем или иным начальным или граничным условиям.

Общие интегралы дифференциальных уравнений и систем

По умолчанию, если дополнительные опции `opts` не указаны, Maple пытается найти явное выражение для неизвестной функции (или функций, если речь идет о системе уравнений). Если не удалось получить явное решение задачи, то решение выводится в виде уравнения, связывающего искомую функцию и независимую переменную, то есть в неявном виде.

В случае, когда решается одно уравнение, функция `dsolve` будет пытаться найти его решение, используя либо методы классификации, либо методы симметрии; в последнем случае функция `dsolve` сначала ищет группы симметрии данного уравнения, а затем использует эту информацию, чтобы проинтегрировать его. Методы классификации используются, когда обыкновенное дифференциальное уравнение соответствует распознаваемой модели, для которой метод решения уже осуществлен. Методы симметрии зарезервированы для неподдающихся классификации случаев.

В Maple предусмотрена возможность проследить за ходом решения при выполнении функции `dsolve`. Для этого перед обращением к функции `dsolve` надо определить системную константу

```
> infolevel[dsolve]:=3:
```

Решения в замкнутой форме (если их удалось найти) возвращаются функцией `dsolve` как последовательность явных уравнений $y(x) = F(x, _Cn)$ или неявного уравнения $F(y(x), x, _Cn) = 0$, где $_Cn$ ($n = \text{integer}$) — произвольные постоянные интегрирования.

Для обыкновенных дифференциальных уравнений первого порядка, особенно, когда они имеют высокую степень в производных dy/dx , решения могут быть выданы в параметрической форме как $[x(_T) = f(_T), y(_T) = g(_T)]$, где $_T$ — параметр, а правые части — явные выражения от $_T$.

Иногда в ответах, возвращаемых функцией `dsolve`, могут появиться интегралы. Эти интегралы выражены с помощью функции инертного интегрирования `Int` и `Intat` (не `int` или `intat`). Они появляются, когда функция `int` не смогла вычислить возникающие при решении интегралы или когда оказалось удобным не вычислять их в процессе решения. Можно запросить вычисление этих интегралов, используя функцию `value`.

Для указания производной при задании уравнения можно использовать функции `diff`, `Diff` или оператор `D`, а для обозначения производной в дополнительных условиях используется только оператор `D`.

Более подробную информацию о функции `dsolve` и используемых параметрах можно получить, сделав запрос `?dsolve`.

В этом разделе мы изучим на примерах применение функции `dsolve` для получения общих интегралов обыкновенных дифференциальных уравнений и систем в аналитической форме.

Задача 2.1

Найти общий интеграл уравнения

$$\frac{dy}{dx} + \frac{xy}{1-x^2} = \arcsin x + x.$$

Решение. Для того чтобы проследить за процессом решения дифференциального уравнения, определим переменную `infolevel`

```
> infolevel[dsolve]:= 3;  
infoleveldsolve := 3
```

Определим уравнение и решим его

```
> eq3:=diff(y(x), x)+x*y(x)/(1-x^2)=arcsin(x)+x;
```

$$eq3 := \left(\frac{d}{dx} y(x) \right) + \frac{x y(x)}{1-x^2} = \arcsin(x) + x$$

```
> res3:=dsolve(eq3, y(x));
```

```
Methods for first order ODEs:
```

```
--- Trying classification methods ---
```

```
trying a quadrature
```

```
trying 1st order linear
```

```
<- 1st order linear successful
```

$$res3 := y(x) = \left(\int \frac{\arcsin(x) + x}{\sqrt{x-1} \sqrt{x+1}} dx + _C1 \right) \sqrt{x-1} \sqrt{x+1}$$

В процессе решения задачи Maple проинформировал нас о ходе решения — были сделаны попытки применения методов классификации для обыкновенных дифференциальных уравнений первого порядка: попытка использования квадратуры; попытка решения линейного дифференциального уравнения 1-ого порядка, которая и завершилась успешно.

Однако в решении присутствует не вычисленный интеграл. Сделаем запрос на вычисление этого интеграла с помощью функции `value`. Для этого мы можем задать наш интеграл, набрав соответствующий код (команду `Int`) на клавиатуре. Однако можно выделить наш интеграл из уже имеющегося выражения `res3`, содержащего этот интеграл.

Рассмотрим, как это можно сделать с помощью команд `nops` и `op`. Как мы знаем, команда `nops` сообщает о количестве объектов первого уровня, а команда `op` выдает их в виде последовательности выражений. Кроме того, при помощи команды `op` можно извлекать подвыражения, указывая номер объекта. В нашем случае будем иметь

```
> nops(res3);
```

```
> op(res3);
```

$$y(x), \left(\int \frac{\arcsin(x) + x}{\sqrt{x-1} \sqrt{x+1}} dx + _C1 \right) \sqrt{x-1} \sqrt{x+1}$$

Итак, Maple-объект `res3` содержит два подобъекта первого уровня. Наш интеграл входит во второй подобъект

```
> op(2, res3);
```

$$\left(\int \frac{\arcsin(x) + x}{\sqrt{x-1} \sqrt{x+1}} dx + _C1 \right) \sqrt{x-1} \sqrt{x+1}$$

причем, как первый подобъект последнего

```
> op(1, %);
```

$$\int \frac{\arcsin(x) + x}{\sqrt{x-1} \sqrt{x+1}} dx + _C1$$

Последняя команда отображает первый подобъект предыдущего результата. В ней в качестве второго параметра используется операция `%`, которая является всего лишь удобной формой ссылки на результат выполнения предыдущей операции.

Выделить наш интеграл из объекта `res3` можно и одной сложносочиненной командой

```
> int_1 := op(1, op(1, rhs(res3)));
```

$$int_1 := \int \frac{\arcsin(x) + x}{\sqrt{x-1} \sqrt{x+1}} dx$$

Теперь попробуем вычислить интеграл с помощью команды `value`

```
> value(int_1);
```

$$\int \frac{\arcsin(x) + x}{\sqrt{x-1} \sqrt{x+1}} dx$$

Неудачно! Подключим студенческий пакет `Calculus1` и запросим подсказку

```
> with(Student[Calculus1]);
```

```
> Hint(int_1);
```

[]

При решении задачи можно попросить подсказку с помощью команды `Hint(expr)`, а также можно попросить сформулировать соответствующее правило с помощью команды `Rule[rule](expr)`. В нашем случае запрос подсказки ничего не дал. Ну что же, такое тоже бывает и достаточно часто. Придется думать самостоятельно. Попробуем представить подынтегральную функцию как сумму двух функций и применить правило суммы для вычисления интеграла

> Rule[rewrite, (arcsin(x)+x)/(x-1)^(1/2)/(x+1)^(1/2) = arcsin(x)/(x^2-1)^(1/2)+x/(x^2-1)^(1/2)](int_1);

$$\int \frac{\arcsin(x)+x}{\sqrt{x-1}\sqrt{x+1}} dx = \int \frac{\arcsin(x)}{\sqrt{-1+x^2}} + \frac{x}{\sqrt{-1+x^2}} dx$$

> Rule[sum](%);

$$\int \frac{\arcsin(x)+x}{\sqrt{x-1}\sqrt{x+1}} dx = \int \frac{\arcsin(x)}{\sqrt{-1+x^2}} dx + \int \frac{x}{\sqrt{-1+x^2}} dx$$

Попробуем теперь вычислить интеграл

> value(%);

$$\int \frac{\arcsin(x)+x}{\sqrt{x-1}\sqrt{x+1}} dx = \int \frac{\arcsin(x)}{\sqrt{-1+x^2}} dx + \sqrt{-1+x^2}$$

Один интеграл взят, но по-прежнему задача не решена до конца. Выделяем не решенный интеграл, при этом учтем, что в силу области допустимых значений $|x| < 1$

> int2:=Int(arcsin(x)/(1-x^2)^(1/2),x)/I;

$$int2 := -I \int \frac{\arcsin(x)}{\sqrt{1-x^2}} dx$$

> value(int2);

$$-\frac{1}{2} I \arcsin(x)^2$$

Теперь все в порядке. Напомним, что в Maple буквой I обозначена мнимая единица. Осталось только собрать полученные результаты в одно целое

> integral:=%+I*(1-x^2)^(1/2);

$$integral := -\frac{1}{2} I \arcsin(x)^2 + \sqrt{1-x^2} I$$

> y:=(integral+C*I)*I*(1-x^2)^(1/2);

$$y := \left(-\frac{1}{2} I \arcsin(x)^2 + \sqrt{1-x^2} I + C I \right) \sqrt{1-x^2} I$$

> simplify(y);

$$\frac{1}{2} (-\arcsin(x)^2 + 2\sqrt{1-x^2} + 2C)\sqrt{1-x^2}$$

> y:=unapply(%,x);

$$y := x \rightarrow \frac{1}{2} (-\arcsin(x)^2 + 2\sqrt{1-x^2} + 2C)\sqrt{1-x^2}$$

Итак, мы получили решение задачи в виде

$$y(x) = \left(\frac{1}{2} \arcsin^2 x - \sqrt{1-x^2} - C \right) \sqrt{1-x^2},$$

где C — произвольная постоянная интегрирования.

Сделаем проверку

```
> simplify(eq3);
```

$$\arcsin(x) + x = \arcsin(x) + x$$

Все в порядке!

Рассмотрим пример решения нелинейного уравнения.

Задача 2.2

Решить уравнение

$$\frac{dy}{dx} + \frac{y}{x} = x^2 y^4.$$

Решение.

```
> y:='y':
```

```
> eq4:=diff(y(x),x)+y(x)/x=x^2*y(x)^4;
```

$$eq4 := \left(\frac{d}{dx} y(x) \right) + \frac{y(x)}{x} = x^2 y(x)^4$$

```
> r4:=dsolve(eq4,y(x));
```

```
Methods for first order ODEs:
```

```
--- Trying classification methods ---
```

```
trying a quadrature
```

```
trying 1st order linear
```

```
trying Bernoulli
```

```
<- Bernoulli successful
```

$$r4 := y(x) = \frac{(-3 \ln(x) - _C1)^2}{(3 \ln(x) - _C1)x},$$

$$y(x) = \frac{\frac{1}{2} \frac{(-3 \ln(x) - _C1)^2}{3 \ln(x) - _C1} - \frac{1}{2} I \sqrt{3} (-3 \ln(x) - _C1)^2}{x},$$

$$y(x) = \frac{\frac{1}{2} \frac{(-3 \ln(x) - _C1)^2}{3 \ln(x) - _C1} + \frac{1}{2} I \sqrt{3} (-3 \ln(x) - _C1)^2}{x}$$

Итак, рассматриваемое уравнение есть уравнение Бернулли. Решение получено в явном виде. Иногда бывает удобнее получить неявное решение; для этого мы используем функцию `dsolve` с опцией `implicit`

```
> y:='y':
> r_4:=dsolve(eq4, y(x), `implicit`);
r_4:=1/y(x)^3 - (-3 ln(x) + _C1)x^3 = 0
```

Таким образом, общий интеграл нашего уравнения определяется соотношением

$$\frac{1}{y^3} - (C_1 - 3 \ln x)x^3 = 0,$$

откуда, считая все величины вещественными, находим

$$y(x) = \frac{1}{x^3 \sqrt[3]{C_1 - 3 \ln x}}$$

или

$$y(x) = \frac{1}{x^3 \sqrt[3]{3 \ln \frac{C}{x}}}.$$

Проверим полученное решение. Для этого используем функцию `odetest`. Функция имеет формат `odetest(sol, ode, y(x))`. Здесь `sol` — проверяемое решение; `ode` — обыкновенное дифференциальное уравнение; `y(x)` — (необязательный параметр) искомая функция.

Функция `odetest` проверяет явные и неявные решения, производя упрощение уравнения относительно данного решения. Если решение имеет место, возвращается "0"; иначе, будет возвращено полученное после упрощения алгебраическое выражение. В случае систем уравнений, функция `odetest` может только проверить явные решения.

Если возвращаемый результат — не ноль, то это не обязательно означает неправильное решение; иногда необходимы некоторые дальнейшие упрощения с использованием функций типа `expand` или `combine`, после чего желанный "0" может быть получен.

Выполним проверку в нашем случае

```
> odetest(r_4, eq4);
```

0

Проверим также явные решения

```
> odetest(r4[1], eq4);
> odetest(r4[2], eq4);
```

```
> odetest(r4[3],eq4);
0
0
0
```

Все в порядке! Рассмотрим еще пример.

Задача 2.3

Решить уравнение

$$3\left(\frac{dy}{dx}\right)^2 = 4y\frac{d^2y}{dx^2} + y^2.$$

Решение.

```
> eq5:=3*diff(y(x),x)^2=4*y(x)*diff(y(x),x$2)+y(x)^2;
```

$$eq5 := 3\left(\frac{d}{dx}y(x)\right)^2 = 4y(x)\left(\frac{d^2}{dx^2}y(x)\right) + y(x)^2$$

```
> r5:=dsolve(eq5,y(x));
```

Methods for second order ODEs:

```
*** Sublevel 2 ***
```

Methods for second order ODEs:

```
--- Trying classification methods ---
```

```
trying 2nd order Liouville
```

```
trying 2nd order, 2 integrating factors of the form
mu(x,y)
```

```
-> Linearization is possible. Trying to determine two
integrating factors as solutions to a 2nd order linear ODE
```

```
-> Tackling the linear ODE "as given":
```

```
trying a quadrature
```

```
checking if the LODE has constant coefficients
```

```
<- constant coefficients successful
```

```
<- successful solving of the linear ODE "as given"
```

$$\text{The integrating factors are: } \left[\frac{\cos\left(\frac{x}{4}\right)}{y^{(3/4)}}, \frac{\sin\left(\frac{x}{4}\right)}{y^{(3/4)}} \right]$$

```
<- 2nd order, 2 integrating factors of the form mu(x,y)
successful
```

```

r5 := y(x) = 0, y(x) = _C1^4 sin(x/4)^4 - 4 _C1^3 sin(x/4)^3 - C2 cos(x/4)
      + 4 cos(x/4) _C1 sin(x/4)^3 - C2^3 - 4 cos(x/4) _C1 sin(x/4) - C2^3 - 6 sin(x/4)^4 - C1^2 - C2^2
      + sin(x/4)^4 - C2^4 - 2 _C2^4 sin(x/4)^2 + 6 _C1^2 sin(x/4)^2 - C2^2 + _C2^4
> odetest(r5[2], eq5);

```

0

Решение не нуждается в дополнительных комментариях. Однако пользоваться таким решением неудобно. Попробуем получить решение в неявной форме

```

> r_5 := dsolve(eq5, y(x), `implicit`);
Methods for second order ODEs:
*** Sublevel 2 ***
Methods for second order ODEs:
--- Trying classification methods ---
trying 2nd order Liouville
trying 2nd order, 2 integrating factors of the form
mu(x, y)
<- 2nd order, 2 integrating factors of the form mu(x, y)
successful

```

$$r_5 := y(x) = 0, -y(x)^{(1/4)} - C_1 \sin\left(\frac{x}{4}\right) + C_2 \cos\left(\frac{x}{4}\right) = 0$$

Сделаем проверку

```

> odetest(r_5[2], eq5);

```

0

Итак, мы получили неявное решение в виде уравнения

$$-y^{1/4} - C_1 \sin\left(\frac{x}{4}\right) + C_2 \cos\left(\frac{x}{4}\right) = 0,$$

откуда легко выделяется и явное решение

```

> isolate(r_5[2], y(x));

```

$$y(x) = \left(-C_1 \sin\left(\frac{x}{4}\right) + C_2 \cos\left(\frac{x}{4}\right)\right)^4$$

Таким образом, окончательно получаем

$$y(x) = \left[C_2 \cos\left(\frac{x}{4}\right) - C_1 \sin\left(\frac{x}{4}\right) \right]^4.$$

Системы обыкновенных дифференциальных уравнений, как мы уже отмечали выше, также решаются с помощью функции `dsolve`. Отличие в применении этой функции для решения системы уравнений состоит в том, что теперь переменная `ode`, задающая систему уравнений, должна быть представлена в виде множества — список уравнений через запятую, заключенный в фигурные скобки. Аналогично, в виде множества, должна быть представлена и переменная `var` — множество определяемых функций.

Задача 2.4

Для системы

$$\frac{dy_1}{dx} = 2y_1 + y_2, \quad \frac{dy_2}{dx} = 2y_2 + y_3, \quad \frac{dy_3}{dx} = 2y_3 + y_1$$

найти общее решение

Решение.

```
> eq1:=diff(y1(x),x)=2*y1(x)+y2(x);
```

```
> eq2:=diff(y2(x),x)=2*y2(x)+y3(x);
```

```
> eq3:=diff(y3(x),x)=2*y3(x)+y1(x);
```

$$eq1 := \frac{d}{dx} y1(x) = 2 y1(x) + y2(x)$$

$$eq2 := \frac{d}{dx} y2(x) = 2 y2(x) + y3(x)$$

$$eq3 := \frac{d}{dx} y3(x) = 2 y3(x) + y1(x)$$

```
> rs1:=dsolve({eq1,eq2,eq3},{y1(x),y2(x),y3(x)});
```

$$rs1 := \{y1(x) = _C1 e^{(3x)} + _C2 e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) + _C3 e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right), y2(x) =$$

$$_C1 e^{(3x)} - \frac{1}{2} _C2 e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) + \frac{1}{2} _C2 e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right) \sqrt{3}$$

$$- \frac{1}{2} _C3 e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right) - \frac{1}{2} _C3 e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) \sqrt{3}, y3(x) = _C1 e^{(3x)}$$

$$- \frac{1}{2} _C2 e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) - \frac{1}{2} _C2 e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right) \sqrt{3} - \frac{1}{2} _C3 e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right)$$

$$+ \frac{1}{2} _C3 e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) \sqrt{3} \}$$

```
> odetest(rs1,{eq1,eq2,eq3});
```

{0}

Здесь мы сразу сделали проверку полученного решения. Итак, мы получили

> assign(rs1);

> y1(x);

$$-C1 e^{(3x)} + C2 e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) + C3 e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right)$$

> y2(x);

$$-C1 e^{(3x)} - \frac{1}{2} C2 e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) + \frac{1}{2} C2 e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right) \sqrt{3} - \frac{1}{2} C3 e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right) - \frac{1}{2} C3 e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) \sqrt{3}$$

> y3(x);

$$-C1 e^{(3x)} - \frac{1}{2} C2 e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) - \frac{1}{2} C2 e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right) \sqrt{3} - \frac{1}{2} C3 e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right) + \frac{1}{2} C3 e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) \sqrt{3}$$

Решения y_2 и y_3 удобно преобразовать, например, собрав вместе константы

> collect(y2(x), [C1, C2, C3]);

$$\left(\frac{1}{2} e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) + \frac{1}{2} e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right) \sqrt{3} \right) C2 + \left(-\frac{1}{2} e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right) - \frac{1}{2} e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) \sqrt{3} \right) C3 + C1 e^{(3x)}$$

> collect(y3(x), [C1, C2, C3]);

$$\left(-\frac{1}{2} e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) - \frac{1}{2} e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right) \sqrt{3} \right) C2 + \left(-\frac{1}{2} e^{\left(\frac{3x}{2}\right)} \cos\left(\frac{\sqrt{3}x}{2}\right) + \frac{1}{2} e^{\left(\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{3}x}{2}\right) \sqrt{3} \right) C3 + C1 e^{(3x)}$$

или, наоборот, собрав вместе синусы, косинусы и экспоненты

> collect(y2(x), [exp, cos, sin]);

$$-C1 e^{(3x)} + \left(\left(-\frac{C2}{2} - \frac{C3\sqrt{3}}{2} \right) \sin\left(\frac{\sqrt{3}x}{2}\right) + \left(\frac{C2\sqrt{3}}{2} - \frac{C3}{2} \right) \cos\left(\frac{\sqrt{3}x}{2}\right) \right) e^{\left(\frac{3x}{2}\right)}$$

> collect(y3(x), [exp, cos, sin]);

$$-C1 e^{(3x)} + \left(\left(-\frac{C2}{2} + \frac{C3\sqrt{3}}{2} \right) \sin\left(\frac{\sqrt{3}x}{2}\right) + \left(-\frac{C2\sqrt{3}}{2} - \frac{C3}{2} \right) \cos\left(\frac{\sqrt{3}x}{2}\right) \right) e^{\left(\frac{3x}{2}\right)}$$

Пакет Maple является незаменимым помощником в обучении различным разделам математики. Это относится и к разделу "Обыкновенные дифференциальные уравнения". Большинство промежуточных шагов решения задачи, используемых функцией `dsolve`, доступно для пользователя, включая все шаги, связанные с применением групп симметрий и интегрирующего множителя. Дополнительные параметры в функции `dsolve` дают возможность решения обыкновенных дифференциальных уравнений с использованием методов, выбранных пользователем, с выполнением или без выполнения шага интегрирования. Все это облегчает понимание, как возможных различных методов решения задачи, так и применения функции `dsolve` и связанных с ней функций Maple, и может успешно использоваться для работы в компьютерном классе.

В Maple имеется специальный пакет DEtools, который в основном предназначен для профессиональной работы с обыкновенными дифференциальными уравнениями. С помощью функций этого пакета можно преобразовывать дифференциальные уравнения с использованием интегрирующих множителей и групп симметрий Ли, понижать порядок рассматриваемого уравнения, вычислять сингулярные точки, преобразовывать уравнения высокого порядка к системе уравнений первого порядка, классифицировать обыкновенные дифференциальные уравнения и многое другое. Большая группа функций этого пакета предназначена для визуализации численных решений задачи Коши для обыкновенных дифференциальных уравнений и систем таких уравнений.

Функции пакета DEtools можно использовать в учебных целях. На этих возможностях пакета мы сейчас кратко и остановимся.

Чтобы получить доступ к функциям пакета DEtools, необходимо подключить его командой

```
> with(DEtools):
```

или загрузить нужную функцию, например, команду `odeadvisor`, при помощи обращения

```
> with(DEtools, odeadvisor):
```

Важную роль в обучении играет функция `odeadvisor`. Основная цель этой функции — классифицировать данное уравнение в соответствии со стандартами, изложенными в справочниках по обыкновенным дифференциальным уравнениям, и отображать справочную страницу, содержащую информацию для решения уравнения. Справочные страницы включают примеры и строки ввода Maple, наряду с некоторым советом, позволяющим пользователю приспособить данную информацию к своей задаче.

Например, рассмотрим уравнение

$$\cos x \frac{dy}{dx} = \frac{y}{\ln y}.$$

```
> eq:=cos(x)*diff(y(x),x)=y(x)/ln(y(x));
```

$$eq := \cos(x) \left(\frac{d}{dx} y(x) \right) = \frac{y(x)}{\ln(y(x))}$$

```
> odeadvisor(eq, help);
[_separable]
```

Таким образом, это — уравнение с разделяющимися переменными. Одновременно с этой информацией, сообщаемой нам на выходе команды `odeadvisor`, открывается окно справочной страницы Maple, в котором мы видим, как следует решать такие уравнения. Если нам не нужна справочная страница Maple, то надо просто использовать функцию `odeadvisor` без опции `help`.

Общий формат функции

```
odeadvisor(ode, y(x), [type1, type2, ...], help)
```

где `ode` — исследуемое уравнение, `y(x)` — неизвестная функция (необязательный параметр), `type1, type2, ...` — (необязательное) подмножество типов классификации обыкновенных дифференциальных уравнений, которые будут проверены, `help` — (необязательный) запрос отображения справочной страницы, основанной на совете учебника для решения заданного уравнения.

Задача 2.5

Исследовать уравнение

$$\cos x \frac{dy}{dx} = y.$$

Решение. Задаем уравнение

```
> eq2:=cos(x)*diff(y(x),x)=y(x);
eq2 := cos(x) \left( \frac{d}{dx} y(x) \right) = y(x)
```

Это, очевидно, линейное уравнение. Проверим

```
> odeadvisor(eq2, [linear]);
[_linear]
```

Решим его как линейное уравнение, причем интеграл не вычисляем

```
> dsolve(eq2, [linear], useInt);
y(x) = _C1 e^{\left( \int \frac{1}{\cos(x)} dx \right)}
```

А теперь найдем окончательное решение, сделав запрос на вычисление интеграла

```
> simplify(value(%));
y(x) = \frac{C1 (1 + \sin(x))}{\cos(x)}
```

Наше уравнение одновременно является и уравнением с разделяющимися переменными, действительно

```
> odeadvisor(eq2);  
[_separable]
```

Решим его теперь как уравнение с разделяющимися переменными, и также пока не будем вычислять интеграл

```
> dsolve(eq2, [separable], useInt);
```

$$\int \frac{1}{\cos(x)} dx - \int \frac{1}{-a} d_a + _CI = 0$$

Теперь попросим вычислить интегралы

```
> simplify(value(%));
```

$$\ln\left(\frac{1 + \sin(x)}{\cos(x)}\right) - \ln(y(x)) + _CI = 0$$

Мы получили решение в неявном виде. Явное решение легко находится с помощью функции

```
> isolate(%, y(x));
```

$$y(x) = -\frac{e^{-CI} \cos(x)}{\sin(x) - 1}$$

Поскольку $_CI$ — произвольная постоянная, то решение можно переписать в виде

$$y(x) = C \frac{\cos x}{\sin x - 1},$$

что, очевидно, совпадает с точностью до постоянного множителя с ранее полученным решением.

Данное уравнение может быть сведено к уравнению в полных дифференциалах. Для этого определим интегрирующий множитель с помощью функции

```
> mu:=intfactor(eq2);
```

$$\mu := \frac{1}{\cos(x) (\sec(x) + \tan(x))}$$

при этом уравнение преобразуется к виду

```
> mu*eq2;
```

$$\frac{\frac{d}{dx} y(x)}{\sec(x) + \tan(x)} = \frac{y(x)}{\cos(x) (\sec(x) + \tan(x))}$$

Решим теперь это уравнение, используя метод exact

```
> dsolve(mu*eq2, [exact]);
```

$$y(x) = -\frac{CI}{\frac{\tan\left(\frac{x}{2}\right) + 1}{2} - 1}$$

Упростим полученное решение

```
> simplify(%);
```

$$y(x) = \frac{-CI \left(\tan\left(\frac{x}{2}\right) + 1 \right)}{-1 + \tan\left(\frac{x}{2}\right)}$$

Рассмотрим еще некоторые возможности пакета `DEtools`. Так, например, команда `varparam` используется для построения общего решения неоднородного уравнения методом вариации произвольных постоянных, если фундаментальная система решений однородного уравнения известна. Эта команда имеет формат вызова `varparam(sols, v, ivar)`; или `DEtools[varparam](sols, v, ivar)`; . Здесь `sols` — список линейно независимых решений линейного однородного дифференциального уравнения; `v` — правая часть неоднородного исходного уравнения; `ivar` — независимая переменная. Может использоваться опция `true`, если необходимо получить только частное решение.

Вот пример использования общей формулы в случае дифференциального уравнения второго порядка:

```
> DEtools[varparam]([u[1](x), u[2](x)], f(x), x);
```

$$-C_1 u_1(x) + C_2 u_2(x) - \int \frac{u_2(x) f(x)}{u_1(x) \left(\frac{d}{dx} u_2(x) \right) - u_2(x) \left(\frac{d}{dx} u_1(x) \right)} dx u_1(x) + \int \frac{u_1(x) f(x)}{u_1(x) \left(\frac{d}{dx} u_2(x) \right) - u_2(x) \left(\frac{d}{dx} u_1(x) \right)} dx u_2(x)$$

```
> DEtools[varparam]([u[1](x), u[2](x)], f(x), x, true);
```

$$- \int \frac{u_2(x) f(x)}{u_1(x) \left(\frac{d}{dx} u_2(x) \right) - u_2(x) \left(\frac{d}{dx} u_1(x) \right)} dx u_1(x) + \int \frac{u_1(x) f(x)}{u_1(x) \left(\frac{d}{dx} u_2(x) \right) - u_2(x) \left(\frac{d}{dx} u_1(x) \right)} dx u_2(x)$$

Задача 2.6

Построить общее решение уравнения

$$\frac{d^2 y}{dx^2} + \lambda y = f(x).$$

Решение. Зададим соответствующее однородное уравнение (левая часть заданного уравнения), найдем фундаментальную систему решений этого однородного уравнения с помощью функции `dsolve` с дополнительной опцией `output=basis`, затем с помощью функции `varparam` построим общее решение нашего неоднородного уравнения с правой частью $f(x)$

```
> deq := diff(y(x), x, x) + lambda*y(x);
> sol := dsolve(deq, y(x), output=basis);
> gsol:=varparam(sol, f(x), x);
```

$$deq := \left(\frac{d^2}{dx^2} y(x) \right) + \lambda y(x)$$

$$sol := [\sin(\sqrt{\lambda} x), \cos(\sqrt{\lambda} x)]$$

$$gsol := _C_1 \sin(\sqrt{\lambda} x) + _C_2 \cos(\sqrt{\lambda} x) + \int \frac{\cos(\sqrt{\lambda} x) f(x)}{\sqrt{\lambda}} dx \sin(\sqrt{\lambda} x)$$

$$- \int \frac{\sin(\sqrt{\lambda} x) f(x)}{\sqrt{\lambda}} dx \cos(\sqrt{\lambda} x)$$

```
> y:=unapply(gsol, x);
```

$$y := x \rightarrow _C_1 \sin(\sqrt{\lambda} x) + _C_2 \cos(\sqrt{\lambda} x) + \int \frac{\cos(\sqrt{\lambda} x) f(x)}{\sqrt{\lambda}} dx \sin(\sqrt{\lambda} x)$$

$$- \int \frac{\sin(\sqrt{\lambda} x) f(x)}{\sqrt{\lambda}} dx \cos(\sqrt{\lambda} x)$$

```
> simplify(deq);
```

$$f(x)$$

Здесь мы создали функцию $y(x)$ на основе полученного решения с помощью команды `unapply` и сразу же проверили это решение с помощью команды `simplify`.

Мы можем задать теперь конкретную функцию, например $f(x) = \sin x$ и получить общее решение нашего уравнения в этом конкретном случае

```
> f:=x->sin(x); value(y(x));
```

$$f := \sin$$

$$\begin{aligned}
& -C_1 \sin(\sqrt{\lambda} x) + -C_2 \cos(\sqrt{\lambda} x) \\
& + \frac{\left(\frac{1}{2} \frac{\cos((\sqrt{\lambda} + 1) x)}{\sqrt{\lambda} + 1} + \frac{1}{2} \frac{\cos((\sqrt{\lambda} - 1) x)}{\sqrt{\lambda} - 1} \right) \sin(\sqrt{\lambda} x)}{\sqrt{\lambda}} \\
& - \frac{\left(\frac{1}{2} \frac{\sin((\sqrt{\lambda} - 1) x)}{\sqrt{\lambda} - 1} - \frac{1}{2} \frac{\sin((\sqrt{\lambda} + 1) x)}{\sqrt{\lambda} + 1} \right) \cos(\sqrt{\lambda} x)}{\sqrt{\lambda}}
\end{aligned}$$

Полученное решение не мешает упростить. Это можно сделать с помощью команд `combine` и `collect`, например, так

```

> combine(value(y(x)));
-C1 sin(sqrt(lambda) x) lambda - C1 sin(sqrt(lambda) x) + -C2 cos(sqrt(lambda) x) lambda - C2 cos(sqrt(lambda) x) + sin(x)
-----
lambda - 1

> collect(%, sin); collect(%, cos);
(-C1 + C1 lambda) sin(sqrt(lambda) x) + sin(x) + -C2 cos(sqrt(lambda) x) lambda - C2 cos(sqrt(lambda) x)
-----
lambda - 1

(C2 lambda - C2) cos(sqrt(lambda) x) + (-C1 + C1 lambda) sin(sqrt(lambda) x) + sin(x)
-----
lambda - 1

> collect(%, _C[1]); collect(%, _C[2]);
-C1 sin(sqrt(lambda) x) + (C2 lambda - C2) cos(sqrt(lambda) x) + sin(x)
-----
lambda - 1

-C2 cos(sqrt(lambda) x) + C1 sin(sqrt(lambda) x) + sin(x)
-----
lambda - 1

```

Итак, мы окончательно получили общее решение задачи в виде

$$y(x) = C_1 \sin(\sqrt{\lambda} x) + C_2 \cos(\sqrt{\lambda} x) + \frac{\sin x}{\lambda - 1}.$$

Впрочем, Maple без труда справляется с этой задачей сразу, непосредственным применением команды `dsolve`

```

> y := 'y': f := 'f':
> deq := diff(y(x), x, x) + lambda*y(x) = f(x);
> sol := dsolve(deq, y(x));
> f := x -> sin(x); dsolve(deq, y(x));

```

$$deq := \left(\frac{d^2}{dx^2} y(x) \right) + \lambda y(x) = f(x)$$

$$\begin{aligned}
sol := y(x) &= \sin(\sqrt{\lambda} x) _C2 + \cos(\sqrt{\lambda} x) _C1 \\
&+ \frac{\int \cos(\sqrt{\lambda} x) f(x) dx \sin(\sqrt{\lambda} x) - \int \sin(\sqrt{\lambda} x) f(x) dx \cos(\sqrt{\lambda} x)}{\sqrt{\lambda}} \\
f &:= \sin \\
y(x) &= \sin(\sqrt{\lambda} x) _C2 + \cos(\sqrt{\lambda} x) _C1 + \frac{\sin(x)}{\lambda - 1}
\end{aligned}$$

Здесь мы рассмотрели лишь небольшую часть возможностей использования функций пакета DEtools. Более полную информацию об этом пакете и примеры можно получить в справочной системе Maple.

Задача Коши и краевая задача для обыкновенных дифференциальных уравнений

Рассмотрим теперь на примерах, как решаются обыкновенные дифференциальные уравнения и системы таких уравнений с дополнительными условиями — начальными или граничными условиями.

Задача 2.7

Найти частный интеграл уравнения

$$\cos x \frac{dy}{dx} = \frac{y}{\ln y},$$

удовлетворяющий начальному условию $y(0) = 1$.

Решение. Зададим уравнение и начальные условия.

```
> eq2:=cos(x)*diff(y(x),x)=y(x)/ln(y(x));
> init2:=y(0)=1;
```

$$\begin{aligned}
eq2 &:= \cos(x) \left(\frac{d}{dx} y(x) \right) = \frac{y(x)}{\ln(y(x))} \\
init2 &:= y(0) = 1
\end{aligned}$$

Данное уравнение есть уравнение с разделяющимися переменными, поэтому при решении нашей задачи сразу укажем тип уравнения

```
> res2:=dsolve({eq2,init2},y(x),[separable]);
```

$$res2 := y(x) = e^{\left(-\sqrt{2} \sqrt{\ln\left(-\frac{\cos(x)}{\sin(x)-1}\right)}\right)}, y(x) = e^{\left(\sqrt{2} \sqrt{\ln\left(-\frac{\cos(x)}{\sin(x)-1}\right)}\right)}$$

Итак, мы получили две интегральные кривые, проходящие через заданную точку $(0, 1)$. Решение задачи имеет вид

$$y(x) = \exp\left(\sqrt{2} \sqrt{\ln\left(\frac{\cos x}{1 - \sin x}\right)}\right), \quad y(x) = \exp\left(-\sqrt{2} \sqrt{\ln\left(\frac{\cos x}{1 - \sin x}\right)}\right).$$

Обратим внимание на то, как теперь используется функция `dsolve` — уравнение и дополнительные условия объединяются в одно множество `{eq2, init2}`, именно данное множество теперь выступает в роли первого параметра функции `dsolve`.

Еще пример — уравнение более высокого порядка.

Задача 2.8

Найти частный интеграл уравнения

$$(x-1) \frac{d^3 y}{dx^3} - \frac{d^2 y}{dx^2} = 0,$$

удовлетворяющий начальным условиям $y(2) = 2$, $y'(2) = 1$, $y''(2) = 1$.

Решение. Зададим уравнение и начальные условия

```
> eq:=(1-x)*(D@@3)(y)(x)-(D@@2)(y)(x)=0;
  init:=y(2)=2,D(y)(2)=1,(D@@2)(y)(2)=1;
      eq:=(1-x)(D^(3))(y)(x)-(D^(2))(y)(x)=0
      init:=y(2)=2,D(y)(2)=1,(D^(2))(y)(2)=1
```

Решаем уравнение

```
> res:=dsolve({eq,init},y(x));
      res:=y(x)=2+(-1+x)ln(-1+x)
```

Проверим найденное решение. Проверить уравнение можно с помощью функции

```
> odetest(res,eq);
```

0

Для проверки начальных условий определим наше решение как функцию от x и упростим каждое из начальных условий


```
> assign(res):
  y:=unapply(y(x),x);
  simplify(value(init[1]));
  simplify(value(init[2]));
  simplify(value(init[3]));
      y:=x->2+(x-1)ln(x-1)
      2=2
      1=1
      1=1
```

Отметим еще одну полезную команду `DEtools[phaseportrait]`. Покажем с помощью этой команды, что, например, решения дифференциального уравнения

$$\frac{dy}{dx} + y = -x^2$$

с различными начальными данными имеют одинаковую асимптотику:

```
> DEtools[phaseportrait](D(y)(x)+y(x)=-x^2, y(x), x=-1..2.5,
[[y(0)=0], [y(0)=1], [y(0)=-1], [y(0)=-2], [y(0)=2]]),
title=`Асимптотика решений`, colour=black, linecolor=black);
```



Решение задачи Коши для системы дифференциальных уравнений ничем не отличается от решения задачи Коши для одного дифференциального уравнения.

Задача 2.9

Проинтегрировать систему уравнений

$$\frac{d^2y}{dx^2} + 2\frac{dz}{dx} - y + 2z = 1, \quad \frac{d^2z}{dx^2} + 2\frac{dy}{dx} + 2y - z = x,$$

и найти решения, удовлетворяющие начальным условиям

$$y(0) = -\frac{7}{9}, \quad y'(0) = \frac{2}{3}, \quad z(0) = -\frac{2}{9}, \quad z'(0) = \frac{1}{3}.$$

Решение. Сначала найдем общее решение системы.

```
> eq1:=diff(y(x), x$2)+2*diff(z(x), x)-y(x)+2*z(x)=1;
```

```
eq2:=diff(z(x), x$2)+2*diff(y(x), x)+2*y(x)-z(x)=x;
```

$$eq1 := \left(\frac{d^2}{dx^2} y(x) \right) + 2 \left(\frac{d}{dx} z(x) \right) - y(x) + 2 z(x) = 1$$

$$eq2 := \left(\frac{d^2}{dx^2} z(x) \right) + 2 \left(\frac{d}{dx} y(x) \right) + 2 y(x) - z(x) = x$$

```

> rs9:=dsolve({eq1,eq2},{y(x),z(x)});
rs9 := {y(x) = -7/9 + 2x/3 + _C2 e^{(-x)} + _C3 e^{(3x)} + _C4 e^{(-x)} x,
z(x) = -_C3 e^{(3x)} + _C4 e^{(-x)} x - 2/9 + x/3 + e^{(-x)} _C1}
> odetest(rs9, {eq1, eq2});
{0}

```

Здесь мы сразу же сделали проверку решения.

Зададим теперь начальные условия и получим частное решение системы

```

> inits:={y(0)=-7/9,D(y)(0)=2/3,z(0)=-2/9,D(z)(0)=1/3};
inits := {y(0) = -7/9, D(y)(0) = 2/3, z(0) = -2/9, D(z)(0) = 1/3}
> rs2:=dsolve({eq1,eq2,inits},{y(x),z(x)});
rs2 := {z(x) = -2/9 + x/3, y(x) = -7/9 + 2x/3}
> odetest(rs2, {eq1, eq2});
{0}

```

Здесь мы сразу проверили выполнение системы уравнений. Для проверки начальных условий определим функции $y(x)$ и $z(x)$ и упростим каждое из начальных условий

```

> assign(rs2):
> y:=unapply(y(x),x);z:=unapply(z(x),x);
y := x → -7/9 + 2/3 x
z := x → -2/9 + 1/3 x
> simplify(inits[1]);simplify(inits[2]);
-7 -7
9 = 9
2 2
3 = 3
-2 -2
9 = 9
1 1
3 = 3

```

Перейдем теперь к изучению на примерах решений граничных задач с помощью Maple.

Задача 2.10

Найти решение уравнения

$$\frac{d^2}{dx^2} \left(g(x) \frac{d^2 y}{dx^2} \right) = f(x)$$

на отрезке $[0, a]$, удовлетворяющее граничным условиям

$$y(0) = 0, y'(0) = 0, y(a) = 0, y'(a) = 0.$$

Решение. Функции $g(x)$ и $f(x)$ будем считать достаточно "хорошими", по крайней мере, такими, что их можно дифференцировать и интегрировать в случае необходимости. Уравнение легко интегрируется в самом общем виде, действительно

> eq4:=Diff(g(x)*Diff(y(x), x\$2), x\$2)=f(x);

$$eq4 := \frac{d^2}{dx^2} \left(g(x) \left(\frac{d^2}{dx^2} y(x) \right) \right) = f(x)$$

> dsolve(eq4, y(x));

$$y(x) = \iint \frac{x_{C2} + _{C1} + \int f(x) dx x - \int x f(x) dx}{g(x)} dx dx + _{C3} x + _{C4}$$

Теперь, задавая конкретный вид функций $g(x)$ и $f(x)$, мы будем получать те или иные общие интегралы уравнения. Если функция $g(x)$ — константа (обозначим эту константу $g(x) = EJ$), а правая часть изменяется, например, по закону синуса

$$f(x) = q_0 \sin\left(\frac{\pi x}{a}\right),$$

то будем иметь

> assign(%):

> f:=x->q0*sin(x*Pi/a);

$$f := x \rightarrow q_0 \sin\left(\frac{x \pi}{a}\right)$$

> g:=x->E*J;

$$g := x \rightarrow EJ$$

> expand(value(y(x)));

$$\frac{q_0 a^4 \sin\left(\frac{\pi x}{a}\right)}{E J \pi^4} + \frac{C_2 x^3}{6 E J} + \frac{C_1 x^2}{2 E J} + C_3 x + C_4$$

Найдем теперь решение нашей граничной задачи при заданных функциях $g(x)$ и $f(x)$. Имея общее решение, можно подставить его в граничные условия и решить полученную систему уравнений относительно произвольных постоянных интегрирования.

Можно получить решение граничной задачи и с помощью функции `dsolve`, как мы это делали в случае задачи Коши. Определим граничные условия

```
> y:='y' ;
> bound:=y(0)=0, D(y)(0)=0, y(a)=0, D(y)(a)=0 ;
bound:=y(0)=0, D(y)(0)=0, y(a)=0, D(y)(a)=0
```

Решим теперь нашу краевую задачу

```
> dsolve({eq4, bound}, y(x));
```

$$y(x) = \frac{q_0 a^4 \sin\left(\frac{x \pi}{a}\right)}{E J \pi^4} + \frac{x^2 a^2 q_0}{E J \pi^3} - \frac{q_0 a^3 x}{E J \pi^3}$$

```
> factor(%);
```

$$y(x) = \frac{q_0 a^2 \left(a^2 \sin\left(\frac{x \pi}{a}\right) + x^2 \pi - a x \pi \right)}{E J \pi^4}$$

Выполним проверку полученного решения. Для этого определим решение как функцию от x и упростим наше уравнение

```
> y:=unapply(rhs(%), x);
```

$$y:=x \rightarrow \frac{q_0 a^2 \left(a^2 \sin\left(\frac{x \pi}{a}\right) + x^2 \pi - a x \pi \right)}{E J \pi^4}$$

```
> simplify(value(eq4));
```

$$q_0 \sin\left(\frac{x \pi}{a}\right) = q_0 \sin\left(\frac{x \pi}{a}\right)$$

Проверим граничные условия

```
> simplify(bound[1]); simplify(bound[2]);
> simplify(bound[3]); simplify(bound[4]);
0=0
0=0
0=0
```

$$0 = 0$$

ПРИМЕЧАНИЕ

Те читатели, кто знаком с сопротивлением материалов, без труда узнают в рассмотренной задаче 2.10 расчет балки с жестко защемленными торцами на синусоидальную нагрузку.

Задача 2.11

Исследовать прогиб балки переменного сечения при постоянной ширине и высоте, меняющейся линейно по длине балки. Рассмотреть два типа закрепления торцов балки: 1) оба торца балки жестко защемлены; 2) один торец балки жестко защемлен, а второй торец — шарнирно оперт.

Решение. Изгиб балки, как известно, описывается следующим дифференциальным уравнением четвертого порядка

$$E \frac{d^2}{dx^2} \left(J(x) \frac{d^2 y}{dx^2} \right) = q(x),$$

где E — модуль Юнга; $J(x)$ — переменный момент инерции поперечного сечения балки; $q(x)$ — заданная поперечная нагрузка. Будем считать, что балка в недеформированном состоянии занимает отрезок $[0, a]$ оси Ox . Граничные условия имеют вид

$$1) \quad y(0) = 0, \quad y'(0) = 0, \quad y(a) = 0, \quad y'(a) = 0 \quad \text{—}$$

условия защемления торцов;

$$2) \quad y(0) = 0, \quad y'(0) = 0, \quad y(a) = 0, \quad y''(a) = 0 \quad \text{—}$$

защемление и шарнирное опирание.

Решаем нашу задачу в Maple. Определяем дифференциальное уравнение изгиба

$$> \text{ode4} := E * \text{Diff} (J(x) * \text{Diff} (y(x), x\$2), x\$2) = q(x);$$

$$\text{ode4} := E \left(\frac{d^2}{dx^2} \left(J(x) \left(\frac{d^2}{dx^2} y(x) \right) \right) \right) = q(x)$$

Определяем коэффициенты уравнения — как известно, момент инерции прямоугольного поперечного сечения с высотой $h(x)$ и шириной b определяется по формуле

$$J(x) = \frac{bh^3(x)}{12}.$$

Высота по условию задачи меняется линейно по длине балки, то есть

$$h(x) = h_0 + \frac{x(h_a - h_0)}{a},$$

где h_0 — высота балки в торце $x = 0$, а h_a — высота балки при $x = a$.

Определяем эти функции в Maple

```
> J:=x->b*h(x)^3/12;h:=x->h0+x*(ha-h0)/a;
```

$$J := x \rightarrow \frac{1}{12} b h(x)^3$$

$$h := x \rightarrow h_0 + \frac{x(ha - h_0)}{a}$$

Конкретизируем теперь нагрузку, действующую на балку. Пусть, например, это будет гидростатическое давление

$$q(x) = q_0 \frac{x}{a}.$$

```
> q:=x->q0*x/a;
```

$$q := x \rightarrow \frac{q_0 x}{a}$$

Выполним расчет, задавшись конкретными числовыми значениями

```
> E:=2e11;b:=0.1;h0:=0.05;ha:=0.1;q0:=10000;a:=5;
```

$$E := 0.2 \cdot 10^{12}$$

$$b := 0.1$$

$$h_0 := 0.05$$

$$ha := 0.1$$

$$q_0 := 10000$$

$$a := 5$$

Посмотрим, какова здесь система базисных функций и найдем общее решение нашего дифференциального уравнения

```
> dsolve(lhs(ode4), output=basis);
```

$$\left[1, x, x^2, \frac{1}{(5+x)^3} \right]$$

```
> dsolve(lhs(ode4));
```

$$y(x) = _C1 + _C2(5+x) + \frac{_C3}{(5+x)^3} + _C4(5+x)^2$$

Решим теперь нашу задачу: вариант граничных условий 1)

```
> ic4_1:=y(0)=0,D(y)(0)=0,y(a)=0,D(y)(a)=0;
```

$$ic4_1 := y(0) = 0, D(y)(0) = 0, y(5) = 0, D(y)(5) = 0$$

```
> res4_1:=evalf(dsolve({ode4,ic4_1},y(x)));
```


$$res4_1 := y(x) = 0.1000000000x^2 - \frac{2.25088500}{5. + x} + 43.05148532 - 26.46968132 \ln(5. + x) - 3.000000000 \ln(5. + x)x + 10.03221458x$$

> assign(res4_1):

> p1:=plot(y(x), x=0..a, color=black, legend='вариант №1'):

Здесь мы получили формулу для вычисления прогиба балки и сразу же подготовили эпюру прогиба.

Вариант граничных условий 2):

> y:='y': ic4_2:=y(0)=0, D(y)(0)=0, y(a)=0, (D@@2)(y)(a)=0;

$$ic4_2 := y(0) = 0, D(y)(0) = 0, y(5) = 0, (D^{(2)})(y)(5) = 0$$

> res4_2:=evalf(dsolve({ode4, ic4_2}, y(x)));

> assign(res4_2):


> p2:=plot(y(x), x=0..5, color=black,

> legend='вариант №2', linestyle=5):

$$res4_2 := y(x) = 0.1000000000x^2 + \frac{5.84754132}{5. + x} + 37.18418860 - 23.83049173 \ln(5. + x) - 3.000000000 \ln(5. + x)x + 9.828313736x$$

Теперь отобразим полученные эпюры прогибов на одном графике

> plots[display]({p1, p2});



Задача 2.12

Исследовать краевую задачу

$$EJ \frac{d^4 y}{dx^4} + P \frac{d^2 y}{dx^2} = 0, \quad EJ = \text{const}, \quad P = \text{const},$$

$$y(0) = 0, \quad y''(0) = 0, \quad y(a) = 0, \quad y'(a) = 0.$$

Решение. Эта задача, очевидно, имеет не представляющее интереса тривиальное решение $y(x) \equiv 0$. Тривиальное решение легко обнаруживается и с помощью Maple. Определим уравнение

```
> ode4 := EJ*Diff(y(x), x$4) + P*Diff(y(x), x$2) = 0;
```

$$\text{ode4} := EJ \left(\frac{d^4}{dx^4} y(x) \right) + P \left(\frac{d^2}{dx^2} y(x) \right) = 0$$

С помощью функции `dsolve` мы можем для линейного уравнения найти базисные функции фундаментального решения дифференциального уравнения, сделав запрос

```
> dsolve(ode4, output=basis);
```

$$\left[1, x, \sin\left(\frac{\sqrt{P} x}{\sqrt{EJ}}\right), \cos\left(\frac{\sqrt{P} x}{\sqrt{EJ}}\right) \right]$$

Из этих базисных функций строится общее решение нашего уравнения.

Определим граничные условия

```
> bound := y(0) = 0, (D@@2)(y)(0) = 0, y(a) = 0, D(y)(a) = 0;
```

$$\text{bound} := y(0) = 0, (D^{(2)})(y)(0) = 0, y(a) = 0, D(y)(a) = 0$$

и попробуем найти решение граничной задачи

```
> dsolve({ode4, bound}, y(x));
```

$$y(x) = 0$$

Как и следовало ожидать, при произвольном значении P мы имеем только тривиальное решение.

Рассматриваемая задача при некоторых значениях P может иметь и ненулевые, нетривиальные решения. Те значения P (точнее, значения параметра P/EJ), для которых существуют нетривиальные решения задачи, называются, как известно, собственными значениями или собственными числами задачи, а соответствующие им нетривиальные решения — собственными функциями задачи.

Попытаемся найти собственные значения и собственные функции нашей задачи. Для этого мы получим сначала общий интеграл нашего уравнения

```
> dsolve(ode4);
```

$$y(x) = _C1 + _C2 x + _C3 \sin\left(\frac{\sqrt{P} x}{\sqrt{EJ}}\right) + _C4 \cos\left(\frac{\sqrt{P} x}{\sqrt{EJ}}\right)$$

Теперь мы подставим найденное общее решение в граничные условия и исследуем полученную систему уравнений относительно произвольных констант. Определим наше решение как функцию

```
> assign(%):y:=unapply(y(x),x);
```

$$y := x \rightarrow _C1 + _C2 x + _C3 \sin\left(\frac{\sqrt{P} x}{\sqrt{EJ}}\right) + _C4 \cos\left(\frac{\sqrt{P} x}{\sqrt{EJ}}\right)$$

Сформируем теперь систему уравнений по граничным условиям

```
> for i from 1 to 4 do e||i:=bound[i] end do;
```

$$e1 := _C1 + _C4 = 0$$

$$e2 := -\frac{C4 P}{EJ} = 0$$

$$e3 := _C1 + _C2 a + _C3 \sin\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) + _C4 \cos\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) = 0$$

$$e4 := _C2 + \frac{_C3 \cos\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) \sqrt{P}}{\sqrt{EJ}} - \frac{_C4 \sin\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) \sqrt{P}}{\sqrt{EJ}} = 0$$

Это — однородная линейная система алгебраических уравнений. Система будет иметь нетривиальное решение, если ее определитель равен нулю. Вычислим определитель полученной системы. Для этого сначала сформируем матрицу системы

```
> A:=linalg[genmatrix]({e1,e2,e3,e4},{_C1,_C2,_C3,_C4});
```

$$A := \begin{bmatrix} 0 & 1 & 1 & 0 \\ 0 & -\frac{P}{EJ} & 0 & 0 \\ \sin\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) & \cos\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) & 1 & a \\ \frac{\cos\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) \sqrt{P}}{\sqrt{EJ}} & -\frac{\sin\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) \sqrt{P}}{\sqrt{EJ}} & 0 & 1 \end{bmatrix}$$

Здесь мы воспользовались полезной командой `genmatrix` из пакета `linalg`, которая формирует матрицу коэффициентов по заданной системе линейных алгебраических уравнений. Формат этой команды очень простой — первым параметром выступает множество уравнений, матрицу коэффициентов которых мы хотим сформировать, а вторым параметром является множество неизвестных, относительно которых рассматривается система уравнений.

Теперь найдем определитель полученной матрицы с помощью команды `det` из пакета `linalg`

```
> Delta:=linalg[det](A);
```

$$\Delta := - \frac{-\sin\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) P EJ^{(3/2)} + \cos\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) P^{(3/2)} a EJ}{EJ^{(5/2)}}$$

Упростим полученное выражение с помощью команды `convert`, преобразуя его к тангенсам

```
> convert(expand(Delta/cos(1/EJ^(1/2)*P^(1/2)*a)),tan);
```

$$\frac{\tan\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) P}{EJ} - \frac{P^{(3/2)} a}{EJ^{(3/2)}}$$

Приравниваем нулю это выражение, то есть, приравниваем нулю наш определитель, и решаем полученное уравнение относительно тангенса

```
> eqP:=isolate(%,tan);
```

$$eqP := \tan\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) = \frac{\sqrt{P} a}{\sqrt{EJ}}$$

Удобно обозначить $\frac{\sqrt{Pa}}{\sqrt{EJ}} = \xi$.

Итак, мы получили следующее характеристическое уравнение $\operatorname{tg} \xi = \xi$

```
> char_P:=tan(xi)=xi;
```

$$char_P := \tan(\xi) = \xi$$

Корни этого уравнения совпадают с нулями нашего определителя, а те, в свою очередь, определяют собственные числа нашей задачи.

Очевидно, характеристическое уравнение имеет бесчисленное множество корней — изолированных точек на вещественной оси. Отрицательные значения ξ , очевидно, не добавляют новых корней характеристического уравнения, поэтому все корни — положительны.

Таким образом, если $\xi = \xi_n$, $n \in N$, где N — множество натуральных чисел, — положительные корни уравнения $\operatorname{tg} \xi = \xi$, то при значениях

$$P = P_n = \frac{\xi_n^2 EJ}{a^2}, \quad n \in N$$

наша задача имеет бесчисленное множество нетривиальных решений — собственных функций.

Множество собственных значений задачи, как известно, называется спектром этой задачи. Таким образом, спектр нашей задачи — счетное множество без точек сгущения. Такой спектр называют дискретным.

Найдем теперь собственные функции задачи

```
> solve({e1,e2},{_C1,_C4});assign(%) :
```

```
> solve({e4},{_C2});
```

$$\left\{ \begin{array}{l} -C2 = -\frac{-C3 \cos\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) \sqrt{P}}{\sqrt{EJ}} \end{array} \right\}$$

```
> assign(%);y(x);
```

$$-\frac{-C3 \cos\left(\frac{\sqrt{P} a}{\sqrt{EJ}}\right) \sqrt{P} x}{\sqrt{EJ}} + -C3 \sin\left(\frac{\sqrt{P} x}{\sqrt{EJ}}\right)$$

Здесь мы из уравнений e1 и e2 нашли константы _C1 и _C4 (они, очевидно, равны нулю), затем из уравнения e4 выразили константу _C2 через _C3. Таким образом, собственные функции будут


$$y = y_n(x) = C \left[\sin\left(\frac{\xi_n x}{a}\right) - \frac{\xi_n x}{a} \cos(\xi_n) \right], n \in N.$$

Напомним, что собственные функции определяются с точностью до произвольного множителя.

На практике часто интерес представляет только первое собственное значение и первая собственная функция. Остальные собственные функции могут быть полезны для построения приближенных решений более сложных задач с теми же граничными условиями. С такими задачами мы еще встретимся в главе, посвященной дифференциальным уравнениям в частных производных. А сейчас определим несколько первых собственных значений и соответствующих им собственных функций нашей задачи.

Посмотрим решение характеристического уравнения на графике

```
> plot({tan(x),x},x=0..12,y=-5..12,color=[black,black],
> linestyle=[3,1],discont=true);
```



Найдем первые три корня характеристического уравнения с помощью команды Roots подпакета Calculus1 пакета Student

```
> xi:=Student[Calculus1][Roots](char_P,xi=0..12,numeric);
      ξ := [4.493409458, 7.725251837, 10.90412166]
```

Посмотрим, как выглядят первые три собственные функции. Для этого определим в Maple собственные функции ($C = 1$)

```
> y:=(x,n)->sin(xi[n]*x/a)-xi[n]*x/a*cos(xi[n]);
```


$$y := (x, n) \rightarrow \sin\left(\frac{\xi_n x}{a}\right) - \frac{\xi_n x \cos(\xi_n)}{a}$$

Нарисуем теперь графики этих функций, задавшись каким-нибудь конкретным значением a , например, $a = 2$

```
> a:=2;
for i from 1 to 3 do
p||i:=plot(y(x,i),x=0..2,color=black,linestyle=2*i-1,
legend=cat(`n = `,convert(i,string))):
end do:
```

$a := 2$

```
> plots[display]({p1,p2,p3});
```



ПРИМЕЧАНИЕ

Те читатели, кто знаком с сопротивлением материалов, без труда узнают в рассмотренной задаче 2.12 расчет прямого стержня с одним жестко защемленным торцом, а другим — шарнирно опертым, на устойчивость при

продольном сжатии. Именно наименьшее (первое) собственное значение $\xi_1 \approx 4,49$ дает критическое значение сжимающей силы

> rhs (eqP) = xi [1] ;

$$\frac{\sqrt{P} a}{\sqrt{EJ}} = 4.493409458$$

> isolate (%, P) ;

$$P = \frac{20.19072856 EJ}{a^2}$$

Собственная функция, соответствующая первому собственному значению, с точностью до масштаба описывает форму изогнутой оси стержня при потере устойчивости.

Функция Грина

Рассмотрим еще один интересный вопрос, связанный с исследованием обыкновенных дифференциальных уравнений, а именно, — построение функции Грина обыкновенных линейных дифференциальных операторов. Знание функции Грина, в частности, позволяет сразу написать частное решение неоднородного обыкновенного линейного дифференциального уравнения, удовлетворяющее заданным граничным условиям.

Мы рассмотрим построение функции Грина для регулярных обыкновенных линейных дифференциальных операторов второго порядка следующего вида

$$Lu \equiv -\frac{d}{dx} \left[p(x) \frac{du}{dx} \right] + q(x) u ,$$

причем, будем считать, что функция $u(x)$ подчинена некоторым заданным граничным условиям. Именно такие операторы наиболее часто встречаются в различных задачах классической математической физики.

Напомним, что функцией Грина краевой задачи называется функция $G(x, \xi)$, удовлетворяющая следующим условиям:

1. при фиксированном $\xi \in [a, b]$ функция $u(x) = G(x, \xi)$ непрерывна; производные $u'(x) = \partial G / \partial x$ и $u''(x) = \partial^2 G / \partial x^2$ также непрерывны в $[a, b]$, кроме точки $x = \xi$, причем скачок функции $u'(x) = \partial G / \partial x$ в точке $x = \xi$ равен

$$\left. \frac{\partial G}{\partial x} \right|_{x=\xi+0} - \left. \frac{\partial G}{\partial x} \right|_{x=\xi-0} = \frac{1}{p(\xi)} ;$$

2. при фиксированном $\xi \in [a, b]$ функция $u(x) = G(x, \xi)$ аргумента x удовлетворяет при $x \neq \xi$, то есть в каждом из интервалов $[a, \xi)$ и $(\xi, b]$, уравнению $Lu = 0$ и заданным краевым условиям при $x = a$ и $x = b$.

Справедливы следующие утверждения:

1. Если краевая задача $Lu = 0$ имеет лишь тривиальное решение, то оператор L имеет одну и только одну функцию Грина;
2. Если уравнение $Lu = 0$ имеет только тривиальное решение, то для любой функции $f(x)$, непрерывной в интервале $[a, b]$, существует решение уравнения $Lu = f$. Это решение задается формулой

$$u(x) = \int_a^b G(x, \xi) u(\xi) d\xi.$$

Последняя формула — формула обращения дифференциального оператора при помощи функции Грина (в этой формуле $G(x, \xi)$ — функция Грина оператора L).

Исходя из сформулированного определения функции Грина, легко построить алгоритм ее нахождения. Приведем соответствующую процедуру

```
> green_function:=proc(LinOp, y, bound1, bound2, G, g1, g2)
  local u, u1, u2, p, L, e1, e2, e3, e4, alpha1, alpha2, beta1, beta2;
  L:=value(LinOp); p:=-coeff(L, diff(y(x), x, x));
  u:=dsolve(L, output=basis);
  u1:=unapply(u[1], x); u2:=unapply(u[2], x);
  alpha1*u1(x)+alpha2*u2(x); g1:=unapply(%, x);
  beta1*u1(x)+beta2*u2(x); g2:=unapply(%, x);
  e1:=value(subs(y=g1, bound1));
  e2:=value(subs(y=g2, bound2));
  e3:=g2(xi)-g1(xi)=0;
  e4:=D(g2)(xi)-D(g1)(xi)=1/subs(x=xi, p);
  solve({e1, e2, e3, e4}, {alpha1, alpha2, beta1, beta2});
  assign(%);
  factor(simplify(expand(g1(x)))); g1:=unapply(-%, x, xi);
  factor(simplify(expand(g2(x)))); g2:=unapply(-%, x, xi);
  G:=(x, xi)->piecewise(x<=xi, g1(x, xi), x>=xi, g2(x, xi));
  G(x, xi);
  G(x, xi):=normal(%); y:='y':
```


end proc:

Процедура имеет семь формальных параметров. Параметр `LinOp` (входной) задает дифференциальное выражение; параметр `y` (входной) — неизвестная функция; параметры `bound1`, `bound2` (входные) — задают граничные условия на левом и правом концах отрезка; параметры `G`, `g1`, `g2` — (выходные) собственно функция Грина и составляющие ее выражения для $x \leq \xi$ и $\xi \leq x$ (эти выражения необходимы для вычисления интеграла при обращении дифференциального оператора).

Рассмотрим применение этой процедуры к решению конкретных задач.

Задача 2.13

Построить функцию Грина следующего оператора

$$-(1+x^2)\frac{d^2y}{dx^2} - 2x\frac{dy}{dx} = 0, \quad y(0) - y'(0) = 0, \quad y(1) = 0.$$

С помощью полученной функции Грина найти решение краевой задачи

$$-(1+x^2)\frac{d^2y}{dx^2} - 2x\frac{dy}{dx} = x^2, \quad y(0) - y'(0) = 0, \quad y(1) = 0.$$

Решение. Определяем нашу задачу в Maple

```
> L := -(1+x^2)*diff(y(x), x, x) - 2*x*diff(y(x), x),
      y(0) - D(y)(0) = 0, y(1) = 0;
```

$$L := -(1+x^2)\left(\frac{d^2}{dx^2}y(x)\right) - 2x\left(\frac{d}{dx}y(x)\right), \quad y(0) - D(y)(0) = 0, \quad y(1) = 0$$

```
> LinOp := L[1]; bound1 := L[2]; bound2 := L[3];
```

$$LinOp := -(1+x^2)\left(\frac{d^2}{dx^2}y(x)\right) - 2x\left(\frac{d}{dx}y(x)\right)$$

$$bound1 := y(0) - D(y)(0) = 0$$

$$bound2 := y(1) = 0$$

Находим функцию Грина с помощью процедуры `grin_function`

```
> green_function(LinOp, y, bound1, bound2, G, g1, g2):
> G(x, xi);
```

$$\begin{cases} \frac{(1 + \arctan(x))(\pi - 4 \arctan(\xi))}{4 + \pi} & x \leq \xi \\ \frac{(1 + \arctan(\xi))(\pi - 4 \arctan(x))}{4 + \pi} & \xi \leq x \end{cases}$$

Построим теперь решение краевой задачи для неоднородного уравнения с помощью найденной функции Грина. Задаем неоднородное уравнение

```
> ode := LinOp = x^2;
```

$$ode := -(1+x^2) \left(\frac{d^2}{dx^2} y(x) \right) - 2x \left(\frac{d}{dx} y(x) \right) = x^2$$

Вычисляем искомую функцию

> `y:=Int(xi^2*g2(x,xi),xi=0..x)+Int(xi^2*g1(x,xi),xi=x..1);`

$$y := \int_0^x \frac{\xi^2 (1 + \arctan(\xi)) (\pi - 4 \arctan(x))}{4 + \pi} d\xi + \int_x^1 \frac{\xi^2 (1 + \arctan(x)) (\pi - 4 \arctan(\xi))}{4 + \pi} d\xi$$

> `y:=unapply(simplify(value(y)),x);`

$$y := x \rightarrow \frac{1}{6}$$

$$\frac{-\pi x^2 + \pi \ln(1+x^2) - 4 \ln(2) + 4 \arctan(x) - 4 \arctan(x) \ln(2) + 4 + 4 \ln(1+x^2) - 4 x^2}{4 + \pi}$$

Сделаем проверку

> `simplify(ode);simplify(bound1);simplify(bound2);`

$$x^2 = x^2$$

$$0 = 0$$

$$0 = 0$$

Как видим, все в порядке!

Для сравнения получим решение нашей краевой задачи непосредственно с помощью функции `dsolve`

> `y:='y':`

> `dsolve({ode,bound1,bound2},y(x));assign(%):`

> `simplify(ode);`

> `y:=unapply(y(x),x);`

> `simplify(bound1);simplify(bound2);`

$$y(x) = -\frac{x^2}{6} + \frac{1}{6} \ln(1+x^2) - \frac{2}{3} \frac{(-1 + \ln(2)) \arctan(x)}{4 + \pi} - \frac{2}{3} \frac{-1 + \ln(2)}{4 + \pi}$$

$$x^2 = x^2$$

$$y := x \rightarrow -\frac{1}{6} x^2 + \frac{1}{6} \ln(1+x^2) - \frac{2}{3} \frac{(-1 + \ln(2)) \arctan(x)}{4 + \pi} - \frac{2}{3} \frac{-1 + \ln(2)}{4 + \pi}$$

$$0 = 0$$

$$0 = 0$$

Здесь мы сразу же сделали проверку найденного решения.

Задача 2.14

С помощью функции Грина найти решение следующей краевой задачи

$$-\frac{d^2 y}{dx^2} - y = e^{-x} \cos x, \quad y(0) = 0, \quad y\left(\frac{\pi}{2}\right) = 0.$$

Решение. Определяем нашу задачу в Maple

```
> _L4 := -diff(y(x), x, x) - y(x), y(0) = 0, y(Pi/2) = 0;
```

$$_L4 := -\left(\frac{d^2}{dx^2} y(x)\right) - y(x), \quad y(0) = 0, \quad y\left(\frac{\pi}{2}\right) = 0$$

Строим функцию Грина

```
> LinOp := _L4[1]; bound1 := _L4[2]; bound2 := _L4[3];
green_function(LinOp, y, bound1, bound2, G, g1, g2):
G(x, xi);
```

$$LinOp := -\left(\frac{d^2}{dx^2} y(x)\right) - y(x)$$

$$bound1 := y(0) = 0$$

$$bound2 := y\left(\frac{\pi}{2}\right) = 0$$

$$\begin{cases} \sin(x) \cos(\xi) & x \leq \xi \\ \sin(\xi) \cos(x) & \xi \leq x \end{cases}$$

Выписываем решение нашей краевой задачи

```
> y := Int(exp(-xi) * cos(xi) * g2(x, xi), xi = 0..x) +
```

```
> Int(exp(-xi) * cos(xi) * g1(x, xi), xi = x..Pi/2);
```

$$y := \int_0^x e^{(-\xi)} \cos(\xi) \sin(\xi) \cos(x) d\xi + \int_x^{\frac{\pi}{2}} e^{(-\xi)} \cos(\xi)^2 \sin(x) d\xi$$

```
> y := combine(value(y));
```

$$y := \frac{2}{5} e^{(-x)} \sin(x) - \frac{1}{5} e^{(-x)} \cos(x) + \frac{1}{5} \cos(x) - \frac{2}{5} e^{\left(-\frac{\pi}{2}\right)} \sin(x)$$

Делаем проверку

```
> y := unapply(y, x);
```

```
> simplify(ode);
```

```
> simplify(bound1);
```

```
> simplify(bound2);
```

$$y := x \rightarrow \frac{2}{5} e^{(-x)} \sin(x) - \frac{1}{5} e^{(-x)} \cos(x) + \frac{1}{5} \cos(x) - \frac{2}{5} e^{\left(-\frac{\pi}{2}\right)} \sin(x)$$

$$e^{(-x)} \cos(x) = e^{(-x)} \cos(x)$$

$$0 = 0$$

$$0 = 0$$

Все в порядке! Получим для сравнения решение краевой задачи непосредственно с помощью функции `dsolve`

```
> y:='y':
dsolve({ode,bound1,bound2},y(x));assign(%);
y:=unapply(y(x),x);
simplify(ode);simplify(bound1);simplify(bound2);
y(x)=-2/5*sin(x)/sqrt(cosh(pi)+sinh(pi))+1/5*cos(x)-1/5*(cos(x)-2*sin(x))*e^(-x)
y:=x->-2/5*sin(x)/sqrt(cosh(pi)+sinh(pi))+1/5*cos(x)-1/5*(cos(x)-2*sin(x))*e^(-x)
e^{(-x)} cos(x) = e^{(-x)} cos(x)
0 = 0
0 = 0
```

Здесь мы также сразу сделали проверку полученного решения. Как мы видим, решения могут быть представлены в самой разной форме.

Линейные дифференциальные уравнения с постоянными коэффициентами

В общей теории обыкновенных дифференциальных уравнений линейные уравнения занимают особое место. Это объясняется тем, что, с одной стороны, они составляют наиболее разработанную часть этой теории, а с другой стороны, описывая реальные физические процессы, линейные уравнения находят многочисленные приложения в физике и технике.

Линейные дифференциальные уравнения с постоянными коэффициентами представляют собой большой и важный класс обыкновенных дифференциальных уравнений, решаемых до конца при помощи элементарных функций.

В этом разделе мы покажем некоторые дополнительные возможности Maple при исследовании таких уравнений и систем. Как мы видели выше, уравнения и системы с постоянными коэффициентами могут быть успешно решены с помощью универсальной функции `dsolve`. Здесь мы рассмотрим примеры решения таких уравнений и систем операционным методом, то есть с помощью преобразования Лапласа.

Как известно, операционный метод особенно эффективен при решении задачи Коши для обыкновенного линейного дифференциального уравнения с постоянными коэффициентами или для системы таких уравнений. Применить операционный метод для решения задачи Коши мы можем двумя способами.

Первый, наиболее простой способ, заключается в том, что указывается соответствующий параметр `opts` в функции `dsolve`, а именно, при обращении к функции `dsolve` необходимо указать `method=laplace`. Следует отметить, что форма представления решения задачи функцией `dsolve` с указанием `method=laplace` и без такого указания — различна, поэтому иногда полезно поэкспериментировать с решением. "Недостатком" такого простого способа применения операционного метода в методическом плане является закрытость процедуры получения решения.

Второй способ состоит в том, что преобразование Лапласа применяется "вручную", то есть, как обычно, сначала получается уравнение (или система уравнений) для трансформанты Лапласа. Затем из этого уравнения находится трансформанта, и, наконец, выполняется обращение преобразования Лапласа. Такой способ решения задачи Коши для обыкновенных дифференциальных уравнений удобно использовать в учебных целях, так как все этапы решения непосредственно видны на рабочем листе. Чтобы воспользоваться этим способом, необходимо подключить пакет интегральных преобразований `inttrans`, имеющийся в Maple. Можно подключить весь пакет с помощью команды

```
> with(inttrans);
[addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert, invlaplace,
 invmellin, laplace, mellin, savetable]
```

или подключить только необходимые функции этого пакета, например, в нашем случае необходимо подключить функцию, выполняющую преобразование Лапласа `laplace` и функцию, выполняющую обратное преобразование Лапласа `invlaplace`

```
> with(inttrans, laplace, invlaplace);
[laplace, invlaplace]
```

Кроме того, можно использовать полную форму обращения, указывая в квадратных скобках имя команды после имени пакета, например,

```
> inttrans[laplace](sin(t)^5, t, p);
120
-----
(p^2 + 1)(p^2 + 9)(p^2 + 25)
```

Рассмотрим применение операционного метода на примерах.

Задача 2.15

Решить следующую задачу Коши

$$\frac{d^2 y}{dt^2} + 5 \frac{dy}{dt} + 6y = 0, \quad y(0) = 0, \quad y'(0) = 1.$$

Решение. Воспользуемся для решения нашей задачи обоими отмеченными выше способами. Сначала определим задачу

```
> ode := diff(y(t), t$2) + 5*diff(y(t), t) + 6*y(t) = 0;
```

```
> init:= y(0) = 0, D(y)(0) = 1;
ode :=  $\left(\frac{d^2}{dt^2} y(t)\right) + 5\left(\frac{d}{dt} y(t)\right) + 6 y(t) = 0$ 
init := y(0) = 0, D(y)(0) = 1
```

Решим задачу операционным методом с использованием функции dsolve

```
> dsolve({ode, init}, y(t), method=laplace);
> assign(%):ode;y:=unapply(y(t), t);init[1];init[2];
```

$$y(t) = 2 e^{\left(-\frac{5t}{2}\right)} \sinh\left(\frac{t}{2}\right)$$

$$0 = 0$$

$$y := t \rightarrow 2 e^{\left(-\frac{5}{2}t\right)} \sinh\left(\frac{1}{2}t\right)$$

$$0 = 0$$

$$1 = 1$$

Здесь мы сразу же выполнили проверку полученного решения.

Итак, решением задачи является функция

$$y(t) = 2e^{-\frac{5}{2}t} \operatorname{sh}\left(\frac{t}{2}\right).$$

Решим теперь нашу задачу "вручную", применяя преобразование Лапласа к заданному уравнению.

Преобразование Лапласа, как мы уже отмечали выше, осуществляется с помощью функции laplace из пакета inttrans. Эта функция имеет следующий формат: laplace(expr, t, p). Здесь expr — выражение, уравнение, или множество выражений и (или) уравнений, которые будут преобразованы; t — переменная выражения expr, по которой выполняется преобразование; p — параметр преобразования.

Выражения включают в себя широкое разнообразие функций: показательные, тригонометрические, функции Бесселя, интегралы вероятности, и многие другие функции, для которых могут быть найдены преобразования Лапласа.

Функция laplace также распознает производные (diff или Diff) и интегралы (int или Int). При преобразовании выражений с производными подобно diff(y(t), t), функция laplace вставляет начальные значения y(0), D(y)(0), и так далее в соответствии с известным свойством преобразования производных (здесь D(y)(0) — значение первой производной в нуле; D(D(y))(0) — второй производной в нуле, и так далее).

Функции laplace и invlaplace распознают дельта-функцию Дирака Dirac(t) и ступенчатую функцию (единичную ступеньку) Хевисайда Heaviside(t).

Функция `invlaplace` имеет точно такой же формат, как и функция `laplace`, а именно, `invlaplace(expr, p, t)` (очевидно, параметры `t` и `p` поменялись местами).

Отметим еще одну полезную возможность — пользователи могут добавлять свои собственные функции к внутренней таблице преобразований Лапласа, используя функцию `addtable`. Более подробную информацию, как всегда, можно получить с помощью запроса `?laplace` или `?addtable`.

Вооружившись необходимыми знаниями относительно функций `laplace` и `invlaplace`, вернемся к нашей задаче. Применяем преобразование Лапласа к заданному уравнению

```
> y:='y':inttrans[laplace](ode, t, p);
      p (p laplace(y(t), t, p) - y(0)) - D(y)(0) + 5 p laplace(y(t), t, p) - 5 y(0)
      + 6 laplace(y(t), t, p) = 0
```

Удобно переобозначить преобразование Лапласа искомой функции $y(t)$, например, как $Y(p)$

```
> subs(init, inttrans[laplace](y(t), t, p) = Y(p), %);
      p^2 Y(p) - 1 + 5 p Y(p) + 6 Y(p) = 0
```

Здесь мы одновременно учли и начальные условия. В результате мы получили уравнение относительно трансформанты Лапласа $Y(p)$. Решаем полученное уравнение

```
> solve(%, Y(p));
```

$$\frac{1}{p^2 + 5p + 6}$$

Находим теперь обратное преобразование Лапласа

```
> inttrans[invlaplace](%, p, t);
```

$$2 e^{\left(-\frac{5t}{2}\right)} \sinh\left(\frac{t}{2}\right)$$

Как и следовало ожидать, мы получили тот же результат!

Для сравнения решим теперь нашу задачу с помощью функции `dsolve`, но при этом не будем указывать метод решения

```
> y:='y':dsolve({ode, init}, y(t));
> assign(%):ode;y:=unapply(y(t), t);init[1];init[2];
```

$$y(t) = -e^{(-3t)} + e^{(-2t)}$$

$$0 = 0$$

$$y := t \rightarrow -e^{(-3t)} + e^{(-2t)}$$

$$0 = 0$$

$$1 = 1$$

Здесь, как обычно, мы сразу же сделали проверку полученного решения. Как мы отмечали выше, теперь решение по форме отличается от ранее полученного операционным методом решения.

Полученные решения достаточно просты и если присмотреться повнимательнее, то можно заметить, что, умножив и разделив последнее решение на 2, а также, вынося за скобки множитель $\exp\left(-\frac{5t}{2}\right)$ и учитывая определение гиперболического синуса, мы приходим к ранее полученному операционным методом решению. В более сложных ситуациях такое совпадение уже не будет столь очевидным; здесь на помощь приходит непосредственная проверка решения.

Задача 2.16

Решить следующую задачу Коши для системы уравнений

$$\frac{d^2x}{dt^2} - 8x + \sqrt{6} \frac{dy}{dt} = 0, \quad \frac{d^2y}{dt^2} + 2y - \sqrt{6} \frac{dx}{dt} = 0,$$

$$x(0) = 1, \quad x'(0) = 0, \quad y(0) = 0, \quad y'(0) = 0.$$

Решение. Определим задачу

$$> \text{ODE1} := \text{diff}(x(t), t\$2) - 8*x(t) + \text{sqrt}(6) * \text{diff}(y(t), t) = 0;$$

$$> \text{ODE2} := -\text{sqrt}(6) * \text{diff}(x(t), t) + \text{diff}(y(t), t\$2) + 2*y(t) = 0;$$

$$\text{ODE1} := \left(\frac{d^2}{dt^2} x(t) \right) - 8 x(t) + \sqrt{6} \left(\frac{d}{dt} y(t) \right) = 0$$

$$\text{ODE2} := -\sqrt{6} \left(\frac{d}{dt} x(t) \right) + \left(\frac{d^2}{dt^2} y(t) \right) + 2 y(t) = 0$$

$$> \text{init} := x(0) = 1, D(x)(0) = 0, y(0) = 0, D(y)(0) = 0;$$

$$\text{init} := x(0) = 1, D(x)(0) = 0, y(0) = 0, D(y)(0) = 0$$

Решаем задачу операционным методом с использованием функции `dsolve`

$$> \text{dsolve}(\{\text{ODE1}, \text{ODE2}, \text{init}\}, \{x(t), y(t)\}, \text{method}=\text{laplace});$$

$$> \text{assign}(\%); x := \text{unapply}(x(t), t); y := \text{unapply}(y(t), t);$$

$$\{y(t) = 8\sqrt{6} \left(-\frac{1}{16} \sin(2t) + \frac{1}{16} \sinh(2t) \right), x(t) = -\frac{1}{2} \cos(2t) + \frac{3}{2} \cosh(2t)\}$$

$$x := t \rightarrow -\frac{1}{2} \cos(2t) + \frac{3}{2} \cosh(2t)$$

$$y := t \rightarrow 8\sqrt{6} \left(-\frac{1}{16} \sin(2t) + \frac{1}{16} \sinh(2t) \right)$$

Проверяем решение


```

> simplify(ODE1);simplify(ODE2);
      0=0
      0=0
> simplify(init[1]);simplify(init[2]);
> simplify(init[3]);simplify(init[4]);
      1=1
      0=0
      0=0
      0=0

```

Все в порядке!

Решим теперь нашу задачу "вручную", применяя преобразование Лапласа к заданной системе уравнений

```

> x:='x':y:='y':
> e1:=inttrans[laplace](ODE1,t,p);
> e2:=inttrans[laplace](ODE2,t,p);
      e1:=p(p laplace(x(t),t,p)-x(0))-D(x)(0)-8 laplace(x(t),t,p)
          +sqrt(6)(p laplace(y(t),t,p)-y(0))=0
      e2:=-sqrt(6)(p laplace(x(t),t,p)-x(0))+p(p laplace(y(t),t,p)-y(0))-D(y)(0)
          +2 laplace(y(t),t,p)=0
> e1:=subs(laplace(x(t),t,p)=X(p),
> laplace(y(t),t,p)=Y(p),init,e1);
> e2:=subs(laplace(x(t),t,p)=X(p),
> laplace(y(t),t,p)=Y(p),init,e2);
      e1:=p(p X(p)-1)-8 X(p)+sqrt(6)p Y(p)=0
      e2:=-sqrt(6)(p X(p)-1)+p^2 Y(p)+2 Y(p)=0

```

Таким образом, мы получили систему алгебраических уравнений относительно трансформант $X(p)$ и $Y(p)$ искомых функций $x(t)$ и $y(t)$. Решаем теперь полученную систему

```

> res:=solve({e1,e2},{X(p),Y(p)});
      res:={X(p)=frac(p(p^2+8),p^4-16),Y(p)=frac(8*sqrt(6),p^4-16)}
> assign(res):X(p);Y(p);
      frac(p(p^2+8),p^4-16)

```

$$\frac{8\sqrt{6}}{p^4 - 16}$$

И, наконец, находим оригиналы

> x:=inttrans[invlaplace](X(p),p,t);

> y:=inttrans[invlaplace](Y(p),p,t);

$$x := \frac{1}{2} \cos(2t) + \frac{3}{2} \cosh(2t)$$

$$y := 8\sqrt{6} \left(\frac{1}{16} \sin(2t) + \frac{1}{16} \sinh(2t) \right)$$

> x:=unapply(x,t);y:=unapply(y,t);

$$x := t \rightarrow \frac{1}{2} \cos(2t) + \frac{3}{2} \cosh(2t)$$

$$y := t \rightarrow 8\sqrt{6} \left(\frac{1}{16} \sin(2t) + \frac{1}{16} \sinh(2t) \right)$$

Так же, как и в задаче 2.15, для сравнения приведем решение с помощью функции dsolve, но при этом не будем указывать метод решения

> x:='x':y:='y':

> dsolve({ODE1,ODE2,init},{x(t),y(t)});

> assign(%);x:=unapply(x(t),t);y:=unapply(y(t),t);

> simplify(ODE1);simplify(ODE2);

> simplify(init[1]);simplify(init[2]);

> simplify(init[3]);simplify(init[4]);

$$\{x(t) = \frac{3}{4} e^{(-2t)} + \frac{3}{4} e^{(2t)} - \frac{1}{2} \cos(2t), y(t) = -\frac{1}{3} \left(\frac{3}{4} e^{(-2t)} - \frac{3}{4} e^{(2t)} + \frac{3}{2} \sin(2t) \right) \sqrt{6} \}$$

$$x := t \rightarrow \frac{3}{4} e^{(-2t)} + \frac{3}{4} e^{(2t)} - \frac{1}{2} \cos(2t)$$

$$y := t \rightarrow -\frac{1}{3} \left(\frac{3}{4} e^{(-2t)} - \frac{3}{4} e^{(2t)} + \frac{3}{2} \sin(2t) \right) \sqrt{6}$$

$$0 = 0$$

$$0 = 0$$

$$1 = 1$$

$$0 = 0$$

$$0 = 0$$

$$0 = 0$$

Проверка показывает, что найденное решение удовлетворяет системе уравнений и всем дополнительным условиям.

Как известно, операционным методом можно найти и общий интеграл уравнения, например, для уравнения задачи 2.15 будем иметь

```
> dsolve(ode, y(t), method=laplace);
```

$$y(t) = e^{\left(-\frac{5t}{2}\right)} \left(y(0) \cosh\left(\frac{t}{2}\right) - (-5 y(0) - 2 D(y)(0)) \sinh\left(\frac{t}{2}\right) \right)$$

```
> collect(%, y(0));
```

$$y(t) = e^{\left(-\frac{5t}{2}\right)} \left(\cosh\left(\frac{t}{2}\right) + 5 \sinh\left(\frac{t}{2}\right) \right) y(0) + 2 e^{\left(-\frac{5t}{2}\right)} \sinh\left(\frac{t}{2}\right) D(y)(0)$$

Таким образом, наше общее решение имеет вид

$$y(t) = e^{-\frac{5}{2}t} \left\{ y(0) \left[\operatorname{ch} \frac{t}{2} + 5 \operatorname{sh} \frac{t}{2} \right] + 2y'(0) \operatorname{sh} \frac{t}{2} \right\}.$$

Здесь начальные данные — произвольные константы.

Аналогичное замечание имеет место и в случае системы дифференциальных уравнений.

Задача Штурма-Лиувилля

В математической физике существенны методы, при которых решение задачи получается в форме ряда или интеграла, то есть в виде разложения по некоторой системе функций. Такие разложения хорошо изучены, когда каждая из функций, по которым осуществляется разложение, зависит только от одной из переменных, встречающихся в задаче. Чтобы найти естественную систему функций, по которой можно осуществить разложение, обычно необходимо найти решение некоторой граничной задачи для обыкновенного дифференциального уравнения, получившей название *задачи Штурма-Лиувилля*.

Поскольку эта задача играет фундаментальную роль при построении решений уравнений в частных производных математической физики, напомним некоторые основные понятия, связанные с ней [3].

Рассмотрим обыкновенное однородное линейное дифференциальное уравнение второго порядка

$$\left[p(x) X'(x) \right]' + [\lambda r(x) - q(x)] X(x) = 0, \quad a < x < b, \quad (2.1)$$

где $p(x)$, $q(x)$, $r(x)$ — вещественные функции от x . Будем предполагать, что $p(x)$, $p'(x)$, $q(x)$, $r(x)$ непрерывны в (a, b) ; $p(x)$ и $r(x)$ положительны в (a, b) ; λ — параметр, принимающий любые значения. Уравнение (2.1) можно записать в виде

$$p(x) X''(x) + p'(x) X'(x) + [\lambda r(x) - q(x)] X(x) = 0 \quad (2.2)$$

или в виде

$$X''(x) + \frac{p'(x)}{p(x)} X'(x) + \left[\lambda \frac{r(x)}{p(x)} - \frac{q(x)}{p(x)} \right] X(x) = 0.$$

Коэффициенты уравнения (2.2) в силу сделанных предположений непрерывны в интервале (a, b) . Каждая внутренняя точка интервала (a, b) — обыкновенная точка уравнения (2.2). Концы интервала (a, b) могут быть как обыкновенными точками, так и особыми (сингулярными). Напомним, что если при некотором x хотя бы один из коэффициентов уравнения (2.2) имеет бесконечный разрыв или $p(x) = 0$, то говорят, что *коэффициенты уравнения имеют особенность* в точке x .

Нас будут интересовать решения уравнения (2.1), удовлетворяющие *однородным линейным граничным условиям с вещественными коэффициентами*. Граничную задачу с такими условиями называют *задачей Штурма–Лиувилля*. Таким образом, под задачей Штурма–Лиувилля понимается следующая задача: найти решения уравнения (2.1), принадлежащие классу $C^{(2)}((a, b))$ и удовлетворяющие некоторым однородным граничным условиям, заданным на концах интервала (a, b) . Примером таких условий могут быть условия

$$X|_{x \rightarrow a+0} = 0, \quad X|_{x \rightarrow b-0} = 0.$$

Различают задачи двух типов — *регулярная задача* и *сингулярная задача*. Задача Штурма–Лиувилля называется *регулярной*, если интервал (a, b) — конечен, концы интервала (a, b) — обыкновенные точки рассматриваемого уравнения. Задача называется *сингулярной*, если хотя бы одно из этих условий не выполнено. Сингулярная задача может быть с одним или двумя сингулярными концами. Характер однородных граничных условий регулярной и сингулярной задач разный.

Сформулируем типовые граничные условия. В случае регулярной задачи различают: *граничные условия первого рода*

$$X(a) = 0, \quad X(b) = 0;$$

граничные условия второго рода

$$X'(a) = 0, \quad X'(b) = 0;$$

граничные условия третьего рода

$$X'(a) - h_a X(a) = 0, \quad X'(b) + h_b X(b) = 0, \quad h_a \geq 0, \quad h_b \geq 0;$$

граничные условия четвертого рода

$$X(a) = X(b), \quad X'(a) = X'(b), \quad p(a) = p(b).$$

Граничные условия четвертого рода называются *условиями периодичности*. Точный смысл сформулированных условий таков

$$X(a) = \lim_{x \rightarrow a+0} X(x), \quad X(b) = \lim_{x \rightarrow b-0} X(x), \quad X'(a) = \lim_{x \rightarrow a+0} X'(x),$$

и так далее. Поэтому они иногда называются предельными условиями.

Все перечисленные условия — *однородные*.

В случае сингулярной задачи различают два варианта задач в зависимости от того, один или два конца сингулярны. Пусть $x = a$ — сингулярный конец, $x = b$ — регулярный конец. Тогда на сингулярном конце ставится условие ограниченности функции

$$X|_{x \rightarrow a+0} = O(1),$$

а на регулярном конце могут быть условия первого, второго или третьего рода, например,

$$X'(b) + h_b X(b) = 0, \quad h_b \geq 0.$$

Если оба конца сингулярны, то ставятся условия ограниченности функции

$$X|_{x \rightarrow a+0} = O(1), \quad X|_{x \rightarrow b-0} = O(1).$$

Перечисленные условия сингулярной задачи также однородны.

Задача Штурма–Лиувилля всегда имеет не представляющее интереса решение $X \equiv 0$, называемое *тривиальным*. Нас будут интересовать *нетривиальные* решения задачи. Однако, нетривиальных решений при данном произвольном λ может и не быть. Поэтому содержанием задачи Штурма–Лиувилля является не только отыскание решений при данном λ , но и определение совокупности значений λ , при которых существуют нетривиальные решения. Всякое нетривиальное решение задачи Штурма–Лиувилля называется *собственной функцией* данной задачи. При этом параметр λ должен принимать некоторые определенные значения, которые называются *собственными значениями* (или *числами*) задачи. Собственные функции по определению находятся с точностью до произвольной константы. Иногда накладывается условие

$$\int_a^b r(x) |X(x)|^2 dx = 1,$$

тогда мы имеем дело с нормированными собственными функциями. Данному собственному значению могут соответствовать одна или две (не более) собственные функции. Множество всех собственных значений называется *спектром* данной задачи.

Отметим некоторые важные свойства собственных значений и собственных функций регулярной задачи Штурма–Лиувилля:

1. Все собственные значения регулярной задачи Штурма–Лиувилля вещественны.
2. Все собственные значения регулярной задачи Штурма–Лиувилля ограничены снизу.
3. Спектр регулярной задачи Штурма–Лиувилля — дискретный, то есть представляет собой счетное множество вещественных чисел без точек сгущения.

4. Система собственных функций регулярной задачи Штурма-Лиувилля ортогональна на отрезке $[a, b]$ с весом $r(x)$.

Второе свойство, в частности, означает, что если

$$m = \inf_{x \in [a, b]} \frac{q(x)}{r(x)},$$

то $\lambda \geq m$. В частности, если $q(x) \equiv 0$, то $\lambda \geq 0$.

Четвертое свойство означает, что если $X_n(x)$ и $X_m(x)$ две собственные функции, отвечающие двум различным собственным значениям λ_n и λ_m , то

$$\int_a^b r(x) X_m(x) X_n(x) dx = \begin{cases} 0, & m \neq n, \\ \|X_n\|^2, & m = n. \end{cases}$$

Пусть теперь некоторая функция $f(x)$ определена на отрезке $[a, b]$. Поставим задачу: представить функцию $f(x)$ в виде ряда

$$f(x) = \sum_{n=1}^{\infty} C_n X_n(x). \quad (2.3)$$

Если предположить, что такое разложение существует и ряд (2.3) сходится так, что он допускает почленное интегрирование, то можно найти коэффициенты разложения C_n

$$C_n = \frac{\int_a^b f(x) r(x) X_n(x) dx}{\|X_n(x)\|^2}.$$

Для обоснования этого разложения на функцию $f(x)$ необходимо наложить определенные условия. Справедлива следующая теорема разложения:

Пусть функция $f(x)$ определена на $[a, b]$ и удовлетворяет там условиям Дирихле:

1. $f(x)$ кусочно-непрерывна на $[a, b]$;
2. $f(x)$ имеет конечное число максимумов и минимумов на $[a, b]$.

Тогда ряд (2.3) сходится к значению функции $f(x)$ во всех точках $x \in [a, b]$, где функция непрерывна. В точке разрыва ряд (2.3) сходится и его сумма равна

$$\frac{f(c+0) + f(c-0)}{2},$$

где $x = c$ — точка разрыва первого рода.

Рассмотрим теперь примеры решения задачи Штурма-Лиувилля в Maple.

Задача 2.17. Найти собственные значения и собственные функции задачи Штурма-Лиувилля:

$$y'' + \lambda y = 0, \quad -y'(a) + hy(a) = 0, \quad -y'(b) + hy(b) = 0, \quad x \in [a, b], \quad h \neq 0.$$

Решение. Определим уравнение в Maple. При этом, рассмотрим два случая $\lambda \neq 0$ и $\lambda = 0$:

```
> eq:=diff(y(x), x$2)+lambda*y(x)=0;
> lambda:=0; eq0:=subs(y(x)=y0(x), eq);
> lambda:='lambda';
```

$$eq := \left(\frac{d^2}{dx^2} y(x) \right) + \lambda y(x) = 0$$

$$\lambda := 0$$

$$eq0 := \frac{d^2}{dx^2} y0(x) = 0$$

$$\lambda := \lambda$$

Рассмотрим сначала случай $\lambda \neq 0$. Найдем общее решение уравнения Штурма-Лиувилля

```
> dsol:=dsolve(eq, y(x)); assign(dsol);
dsol := y(x) = _C1 sin(sqrt(lambda) x) + _C2 cos(sqrt(lambda) x)
```

Удобно переписать это общее решение в другой форме

```
> y := x->
C1*sin(lambda^(1/2)*(x-a))+C2*cos(lambda^(1/2)*(x-a));
simplify(value(eq));
y := x -> C1 sin(sqrt(lambda) (x - a)) + C2 cos(sqrt(lambda) (x - a))
0 = 0
```

Здесь мы сразу же и проверили справедливость такого представления решения.

Для дальнейших вычислений нам понадобится производная решения, которую мы определим так

```
> y1:=D(y);
y1 := x -> C1 cos(sqrt(lambda) (x - a)) sqrt(lambda) - C2 sin(sqrt(lambda) (x - a)) sqrt(lambda)
```

Подставим теперь найденное общее решение в граничные условия

```
> eq1:=-y1(a)+h*y(a)=0;
> eq2:=-y1(b)+h*y(b)=0;
```

$$\begin{aligned}
 eq1 &:= -C1 \sqrt{\lambda} + h C2 = 0 \\
 eq2 &:= -C1 \cos(\sqrt{\lambda} (b - a)) \sqrt{\lambda} + C2 \sin(\sqrt{\lambda} (b - a)) \sqrt{\lambda} \\
 &\quad + h (C1 \sin(\sqrt{\lambda} (b - a)) + C2 \cos(\sqrt{\lambda} (b - a))) = 0
 \end{aligned}$$

Таким образом, мы имеем однородную линейную систему алгебраических уравнений для определения произвольных постоянных интегрирования C_1 и C_2 . Для того, чтобы эта система имела нетривиальное решение, необходимо и достаточно, чтобы ее определитель был равен нулю. Поэтому найдем определитель системы. Для этого сформируем матрицу системы и вычислим ее определитель

```

> A:=linalg[genmatrix]({eq1,eq2},{C1,C2}):
> Delta:=simplify(linalg[det](A));

$$\Delta := \sin(\sqrt{\lambda} (-b + a)) (\lambda + h^2)$$


```

Здесь мы воспользовались командами `genmatrix` и `det` из пакета `linalg`. Первая — формирует матрицу системы уравнений $\{eq1, eq2\}$ относительно неизвестных $\{C1, C2\}$. Вторая команда вычисляет определитель этой матрицы. Приравниваем нулю этот определитель и находим множество собственных значений

```

> _EnvAllSolutions:=true;
> solve(Delta=0,sqrt(lambda));

$$\_EnvAllSolutions := true$$


$$\frac{\pi \_Z1\sim}{-b + a}$$


```

Чтобы получить все множество решений тригонометрического уравнения, мы определили системную переменную как `_EnvAllSolutions:=true`. Результат решения содержит целочисленную системную переменную `_Z1~`, в которой знак тильда означает, что на значения переменной наложены некоторые ограничения. Таким образом, собственные значения будут

$$\lambda_n = \frac{\pi^2 n^2}{(b - a)^2},$$

или, в Maple

```

> lambda:=subs(_Z1=n,%^2);

$$\lambda := \frac{\pi^2 n^2}{(-b + a)^2}$$

> lambda[n]:=%;

$$\lambda_n := \frac{\pi^2 n^2}{(-b + a)^2}$$


```

Здесь мы ввели привычное обозначение n для целочисленной переменной.

Определим теперь соответствующие собственные функции:

> C2:=solve(eq1,C2);

$$C2 := \frac{C1 \sqrt{\lambda}}{h}$$

> collect(y(x),C1);

$$\left(\sin(\sqrt{\lambda} (x - a)) + \frac{\sqrt{\lambda} \cos(\sqrt{\lambda} (x - a))}{h} \right) C1$$

> subs(lambda=lambda[n],select(has,%,x));

$$\sin\left(\sqrt{\frac{\pi^2 n^2}{(-b+a)^2}} (x - a)\right) + \frac{\sqrt{\frac{\pi^2 n^2}{(-b+a)^2}} \cos\left(\sqrt{\frac{\pi^2 n^2}{(-b+a)^2}} (x - a)\right)}{h}$$

> assume(n, posint):assume(b>a):

Y:=collect(simplify(%),sin);

$$Y := -\frac{(h b - h a) \sin\left(\frac{\pi n (-x + a)}{-b + a}\right)}{(-b + a) h} - \frac{\pi n \cos\left(\frac{\pi n (-x + a)}{-b + a}\right)}{(-b + a) h}$$

> simplify(op(1,Y));

$$\sin\left(\frac{\pi n (-x + a)}{-b + a}\right)$$

> y:=unapply(%+op(2,Y),x,n);

$$y := (x, n) \rightarrow \sin\left(\frac{\pi n (-x + a)}{-b + a}\right) - \frac{\pi n \cos\left(\frac{\pi n (-x + a)}{-b + a}\right)}{(-b + a) h}$$

Вычисления не нуждаются в комментариях, отметим только, что пришлось выполнить (с помощью Maple) некоторые упрощения. Итак, собственные функции будут

$$y_n(x) = \sin\left(\frac{n\pi(x-a)}{b-a}\right) + \frac{n\pi}{h(b-a)} \cos\left(\frac{n\pi(x-a)}{b-a}\right).$$

Проверим ортогональность собственных функций на отрезке $[a, b]$:

> Int(y(x,n)*y(x,m),x=a..b);res:=value(%);

$$\int_a^b \left(\sin\left(\frac{\pi n (-x + a)}{-b + a}\right) - \frac{\pi n \cos\left(\frac{\pi n (-x + a)}{-b + a}\right)}{(-b + a) h} \right) \left(\sin\left(\frac{\pi m (-x + a)}{-b + a}\right) - \frac{\pi m \cos\left(\frac{\pi m (-x + a)}{-b + a}\right)}{(-b + a) h} \right) dx$$

$$res := \frac{(-1)^{n\sim} \sin(\pi m) n\sim (h^2 a\sim^2 - 2 h^2 b\sim a\sim + \pi^2 m^2 + h^2 b\sim^2)}{\pi h^2 (-n\sim^2 b\sim + n\sim^2 a\sim + m^2 b\sim - m^2 a\sim)}$$

> assume(m, posint) : simplify(res) ;

0

Все в порядке! Вычислим теперь квадрат нормы этих собственных функций:

> Int(y(x, n)^2, x=a..b) ; Norma := simplify(value(%)) ;

$$\int_{a\sim}^{b\sim} \left(\sin\left(\frac{\pi n\sim (-x + a\sim)}{-b\sim + a\sim}\right) - \frac{\pi n\sim \cos\left(\frac{\pi n\sim (-x + a\sim)}{-b\sim + a\sim}\right)}{(-b\sim + a\sim) h} \right)^2 dx$$

$$Norma := - \frac{h^2 a\sim^2 - 2 h^2 b\sim a\sim + \pi^2 n\sim^2 + h^2 b\sim^2}{2 (-b\sim + a\sim) h^2}$$

> Norma := collect(Norma, h) ;

$$Norma := - \frac{b\sim^2 - 2 b\sim a\sim + a\sim^2}{2 (-b\sim + a\sim)} - \frac{\pi^2 n\sim^2}{2 (-b\sim + a\sim) h^2}$$

> simplify(op(1, Norma)) ;

$$\frac{b\sim}{2} - \frac{a\sim}{2}$$

> Norma := % + op(2, Norma) ;

$$Norma := \frac{b\sim}{2} - \frac{a\sim}{2} - \frac{\pi^2 n\sim^2}{2 (-b\sim + a\sim) h^2}$$

Итак,

$$\|y_n\|^2 = \frac{b-a}{2} + \frac{n^2 \pi^2}{2h^2(b-a)}.$$

Проверим теперь, будет ли число $\lambda_0 = 0$ собственным значением нашей задачи:

> y0 := 'y0' ;

> sol0 := dsolve(eq0, y0(x)) ; assign(sol0) ;

> simplify(value(eq0)) ; y0 := unapply(y0(x), x) ;

$$sol0 := y0(x) = _C1 x + _C2$$

$$0 = 0$$

$$y0 := x \rightarrow _C1 x + _C2$$

> eq1_0 := -D(y0)(a) + h*y0(a) = 0 ; eq2_0 := -D(y0)(b) + h*y0(b) = 0 ;

$$eq1_0 := -_C1 + h(_C1 a\sim + _C2) = 0$$

```

eq2_0 := -_C1 + h (_C1 b~ + _C2) = 0
> A0:=linalg[genmatrix]({eq1_0,eq2_0},{_C1,_C2});
> Delta0:=simplify(linalg[det](A0));

```

$$A0 := \begin{bmatrix} -1 + h a\sim & h \\ -1 + h b\sim & h \end{bmatrix}$$

$$\Delta 0 := -h^2 b\sim + h^2 a\sim$$

Видим, что определитель не равен нулю, следовательно, однородная система имеет только тривиальное решение. Значит, число $\lambda_0 = 0$ не является собственным значением нашей задачи.

Рассмотрим еще одну похожую задачу.

Задача 2.18. Найти собственные значения и собственные функции задачи Штурма-Лиувилля:

$$y'' + \lambda y = 0, \quad -y'(0) + hy(0) = 0, \quad y'(a) + hy(a) = 0, \quad x \in [0, a], \quad h > 0.$$

Решение. Определим задачу в Maple. Опять рассмотрим два случая $\lambda \neq 0$ и $\lambda = 0$:

```

> eq:=diff(y(x),x$2)+lambda*y(x)=0;
> lambda:=0;eq0:=subs(y(x)=y0(x),eq);
> lambda:='lambda';

```

$$eq := \left(\frac{d^2}{dx^2} y(x) \right) + \lambda y(x) = 0$$

$$\lambda := 0$$

$$eq0 := \frac{d^2}{dx^2} y0(x) = 0$$

$$\lambda := \lambda$$

Пусть $\lambda \neq 0$. Находим общее решение уравнения и создаем соответствующую функцию

```

> dsol:=dsolve(eq,y(x));assign(dsol);
      dsol := y(x) = _C1 sin(sqrt(lambda) x) + _C2 cos(sqrt(lambda) x)
> y:=unapply(y(x),x);#simplify(value(eq));
      y := x -> _C1 sin(sqrt(lambda) x) + _C2 cos(sqrt(lambda) x)

```

Вычисляем производную

```

> y1:=D(y);
      y1 := x -> _C1 cos(sqrt(lambda) x) sqrt(lambda) - _C2 sin(sqrt(lambda) x) sqrt(lambda)

```

Формируем систему уравнений по граничным условиям и вычисляем определитель этой системы

```

> eq1:=-y1(0)+h*y(0)=0;
eq2:=y1(a)+h*y(a)=0;
      eq1 := -_C1 sqrt(lambda) + h _C2 = 0
      eq2 := _C1 cos(sqrt(lambda) a) sqrt(lambda) - _C2 sin(sqrt(lambda) a) sqrt(lambda)
              + h (_C1 sin(sqrt(lambda) a) + _C2 cos(sqrt(lambda) a)) = 0
> A:=linalg[genmatrix]({eq1,eq2},{_C1,_C2});
Delta:=simplify(linalg[det](A));
      A := [ [-sqrt(lambda) h]
             [cos(sqrt(lambda) a) sqrt(lambda) + h sin(sqrt(lambda) a) -sin(sqrt(lambda) a) sqrt(lambda) + h cos(sqrt(lambda) a)] ]
      Delta := sin(sqrt(lambda) a) lambda - 2 sqrt(lambda) h cos(sqrt(lambda) a) - h^2 sin(sqrt(lambda) a)
> eq:=expand(simplify(Delta/cos(lambda^(1/2)*a)))=0;
      eq := (sin(sqrt(lambda) a) lambda) / cos(sqrt(lambda) a) - 2 sqrt(lambda) h - (h^2 sin(sqrt(lambda) a)) / cos(sqrt(lambda) a) = 0

```

В результате приравнивания нулю определителя мы получили характеристическое уравнение eq для определения собственных значений. Удобно преобразовать это уравнение, перейдя к тангенсам

```

> convert(eq,tan);
      tan(sqrt(lambda) a) lambda - 2 sqrt(lambda) h - h^2 tan(sqrt(lambda) a) = 0
> eq:=collect(%,tan);
      eq := (lambda - h^2) tan(sqrt(lambda) a) - 2 sqrt(lambda) h = 0

```

Удобно ввести новую переменную по формуле $\lambda = \left(\frac{\mu}{a}\right)^2$

```

> assume(mu>0):assume(a>0);
> simplify(subs(lambda=(mu/a)^2,eq));
      (tan(mu~) mu~^2 - tan(mu~) h^2 a~^2 - 2 mu~ h a~) / a~^2 = 0
> eq:=isolate(%,tan(mu));
      eq := tan(mu~) = (2 mu~ h a~) / (mu~^2 - h^2 a~^2)

```

Согласно общей теории все собственные значения ограничены снизу, в данном случае $\lambda > 0$. Пусть $\mu_n, n = 1, 2, \dots$ — положительные корни уравнения

$$\operatorname{tg} \mu = \frac{2\mu h a}{\mu^2 - h^2 a^2}.$$

Тогда собственные значения нашей задачи будут $\lambda_n = \left(\frac{\mu_n}{a}\right)^2$, $n = 1, 2, \dots$

Найдем теперь соответствующие собственные функции

```
> _C2:=solve(eq1,_C2);
_C2:=simplify(subs(lambda=(mu/a)^2,_C2));
_C2 :=  $\frac{C1 \sqrt{\lambda}}{h}$ 
_C2 :=  $\frac{C1 \mu}{a h}$ 
> collect(simplify(subs(lambda=(mu/a)^2,y(x))),_C1);

$$\frac{-C1 \left( \sin\left(\frac{\mu x}{a}\right) a h + \mu \cos\left(\frac{\mu x}{a}\right) \right)}{a h}$$

> y:=unapply(subs(mu=mu[n],select(has,%,x)),x,n);
y := (x, n) →  $\sin\left(\frac{\mu_n x}{a}\right) a h + \mu_n \cos\left(\frac{\mu_n x}{a}\right)$ 
```

Итак, собственные функции имеют вид

$$y_n(x) = ah \sin\left(\frac{\mu_n x}{a}\right) + \mu_n \cos\left(\frac{\mu_n x}{a}\right).$$

Проверим ортогональность этих функций на отрезке $[0, a]$

```
> Int(y(x,n)*y(x,m),x=0..a);res:=value(%);

$$\int_0^a \left( \sin\left(\frac{\mu_n x}{a}\right) a h + \mu_n \cos\left(\frac{\mu_n x}{a}\right) \right) \left( \sin\left(\frac{\mu_m x}{a}\right) a h + \mu_m \cos\left(\frac{\mu_m x}{a}\right) \right) dx$$

res :=  $-a (h^2 a^2 \mu_n \cos(\mu_n) \sin(\mu_m) + \mu_n \mu_m^2 \cos(\mu_n) \sin(\mu_m) - h^2 a^2 \mu_m \sin(\mu_n) \cos(\mu_m) - a h \mu_n^2 \sin(\mu_n) \sin(\mu_m) - \mu_n^2 \mu_m \sin(\mu_n) \cos(\mu_m) + a h \mu_m^2 \sin(\mu_n) \sin(\mu_m)) / (\mu_n^2 - \mu_m^2)$ 
> simplify(res,
{(mu[n]^2-h^2*a^2)*sin(mu[n])=2*mu[n]*h*a*cos(mu[n]),
(mu[m]^2-h^2*a^2)*sin(mu[m])=2*mu[m]*h*a*cos(mu[m])});
```

Все в порядке! Вычислим квадрат нормы

```
> Int (y(x, n) ^2, x=0..a);
Norma:=simplify(value(%),
{ (mu[n]^2-h^2*a^2)*sin(mu[n])=2*mu[n]*h*a*cos(mu[n]) });
```

$$\int_0^a \left(\sin\left(\frac{\mu_n x}{a}\right) a h + \mu_n \cos\left(\frac{\mu_n x}{a}\right) \right)^2 dx$$

$$Norma := \frac{1}{2} \frac{h^2 a^3 \mu_n + 2 \mu_n a^2 h + \mu_n^3 a}{\mu_n}$$

```
> Norma:=factor(expand(Norma));
```

$$Norma := \frac{1}{2} a (h^2 a^2 + 2 h a + \mu_n^2)$$

Итак, квадрат нормы равен

$$\|y_n\|^2 = \frac{a(h^2 a^2 + 2 h a + \mu_n^2)}{2}.$$

Проверим теперь число $\lambda = 0$.

```
> sol0:=dsolve(eq0, y0(x));
> assign(sol0);simplify(value(eq0));
sol0 := y0(x) = _C1 x + _C3
0 = 0
> y1_0:=diff(y0(x), x);
y1_0 := _C1
> eq1_0:=-y1_0+h*subs(x=0, y0(x));
> eq2_0:=y1_0+h*subs(x=a, y0(x));
eq1_0 := -_C1 + h _C3
eq2_0 := _C1 + h (_C1 a + _C3)
> A0:=linalg[genmatrix]({eq1_0,eq2_0},{_C1,_C3});
> Delta0:=simplify(linalg[det](A0));
```

$$A0 := \begin{bmatrix} h & -1 \\ h & 1 + a h \end{bmatrix}$$

$$\Delta 0 := 2 h + a h^2$$

Очевидно, определитель не равен нулю, значит, система уравнений имеет только тривиальное решение. Число $\lambda = 0$ не является собственным значением задачи.

Задача 2.19. Разложить функцию $f(x) = 1 - \frac{x^2}{b^2}$ на отрезке $[0, b]$ в ряд по собственным функциям задачи Штурма-Лиувилля:

$$\frac{d}{dx} \left(x \frac{dy}{dx} \right) + \lambda xy = 0, \quad y(0) = O(1), \quad y(b) = 0.$$

Решение. В данном случае мы имеем сингулярную задачу Штурма-Лиувилля (точка $x = 0$ — особая точка уравнения). Если спектр задачи дискретный, то теория регулярной задачи переносится на случай сингулярной задачи. Определим задачу в Maple:

```
> eq:=diff(x*diff(y(x),x),x)+lambda*x*y(x)=0;
```

$$eq := \left(\frac{d}{dx} y(x) \right) + x \left(\frac{d^2}{dx^2} y(x) \right) + \lambda x y(x) = 0$$

Находим общее решение

```
> res:=dsolve(eq,y(x));
```

$$res := y(x) = _C1 \text{ BesselJ}(0, \sqrt{\lambda} x) + _C2 \text{ BesselY}(0, \sqrt{\lambda} x)$$

Наше уравнение является уравнением Бесселя с нулевым индексом. Общее решение этого уравнения выражается через две линейно независимые функции Бесселя первого и второго рода — собственно функцию Бесселя $J_0(\sqrt{\lambda}x)$ (функция первого рода) и функцию Вебера $Y_0(\sqrt{\lambda}x)$ (функция второго рода).

Как известно, ограниченной в нуле является только функция Бесселя первого рода.

Поэтому, в силу граничных условий, мы должны положить в общем решении

$_C2 = 0$:

```
> subs(_C2=0,rhs(res));
```

$$_C1 \text{ BesselJ}(0, \sqrt{\lambda} x)$$

А тогда из второго граничного условия находим ($_C1 \neq 0$)

```
> eq1:=subs(x=b,%)/_C1=0;
```

$$eq1 := \text{BesselJ}(0, \sqrt{\lambda} b) = 0$$

Это и есть уравнение для определения собственных значений задачи. Удобно обозначить $\alpha = \sqrt{\lambda}b$. Тогда характеристическое уравнение будет

```
> eq1:=subs(lambda^(1/2)*x=alpha,eq1);
```

$$eq1 := \text{BesselJ}(0, \alpha) = 0$$

Пусть $\alpha_n, n = 1, 2, \dots$ — положительные корни уравнения $J_0(\alpha) = 0$ (то есть, уравнения eq1). Тогда собственные значения задачи будут $\lambda_n = \left(\frac{\alpha_n}{b}\right)^2, n = 1, 2, \dots$

Соответствующие собственные функции будут $y_n(x) = J_0\left(\frac{\alpha_n x}{b}\right)$:

```
> Yn := (x, n) -> BesselJ(0, alpha[n] * x/b);
```

$$Y_n := (x, n) \rightarrow \text{BesselJ}\left(0, \frac{\alpha_n x}{b}\right)$$

Проверим дифференциальное уравнение

```
> y := 'y' : Yn(x, n) :
```

```
> simplify(subs(lambda = (alpha[n]/b)^2, y(x) = %, eq1));
```

$$0 = 0$$

Проверим граничное условие

```
> subs(alpha = alpha[n], eq1); simplify(Yn(b, n), {%}) = 0;
```

$$\text{BesselJ}(0, \alpha_n) = 0$$

$$0 = 0$$

Все в порядке!

Проверим ортогональность собственных функций на отрезке $[0, b]$ с весом x :

```
> assume(n, posint) : assume(m, posint) :
```

```
int(x * Yn(x, n) * Yn(x, m), x = 0..b); intJ := simplify(value(%));
```

$$\int_0^b x \text{BesselJ}\left(0, \frac{\alpha_{n-} x}{b}\right) \text{BesselJ}\left(0, \frac{\alpha_{m-} x}{b}\right) dx$$

$$\text{intJ} := \frac{b^2 (\alpha_{m-} \text{BesselJ}(0, \alpha_{n-}) \text{BesselJ}(1, \alpha_{m-}) - \alpha_{n-} \text{BesselJ}(1, \alpha_{n-}) \text{BesselJ}(0, \alpha_{m-}))}{-\alpha_{n-}^2 + \alpha_{m-}^2}$$

```
> e1 := subs(alpha = alpha[n], eq1);
```

```
> e2 := subs(alpha = alpha[m], eq1);
```

```
> simplify(intJ, {e1, e2});
```

$$e1 := \text{BesselJ}(0, \alpha_{n-}) = 0$$

$$e2 := \text{BesselJ}(0, \alpha_{m-}) = 0$$

$$0$$

Все в порядке!

Вычислим теперь квадрат нормы

> Norma := Int (x*Yn (x, n) ^2, x=0..b) ;

> Norma := simplify (value (Norma) , {e1}) ;

$$Norma := \int_0^b x \text{BesselJ} \left(0, \frac{\alpha_n x}{b} \right)^2 dx$$

$$Norma := \frac{1}{2} b^2 \text{BesselJ} (1, \alpha_n)^2$$

Итак, мы получили известную формулу [3]

$$\left\| J_0 \left(\frac{\alpha_n x}{b} \right) \right\|^2 \equiv \int_0^b x J_0^2 \left(\frac{\alpha_n x}{b} \right) dx = \frac{b^2}{2} J_1^2 (\alpha_n).$$

Приступим теперь к разложению заданной функции в ряд. Определим нашу функцию в Maple

> f := x -> 1 - x^2/b^2 ;

$$f := x \rightarrow 1 - \frac{x^2}{b^2}$$

Коэффициенты разложения, как известно, вычисляются по формуле

$$C_n = \frac{1}{\left\| J_0 \left(\frac{\alpha_n x}{b} \right) \right\|^2} \int_0^b x \left(1 - \frac{x^2}{b^2} \right) J_0 \left(\frac{\alpha_n x}{b} \right) dx \Rightarrow$$

$$C_n = \frac{2}{b^2 J_1^2 (\alpha_n)} \int_0^b x \left(1 - \frac{x^2}{b^2} \right) J_0 \left(\frac{\alpha_n x}{b} \right) dx.$$

> Cn := Int (x*f (x) *Yn (x, n) , x=0..b) /Norma ;

$$Cn := \frac{2}{b^2 \text{BesselJ} (1, \alpha_n)^2} \int_0^b x \left(1 - \frac{x^2}{b^2} \right) \text{BesselJ} \left(0, \frac{\alpha_n x}{b} \right) dx$$

> Cn := simplify (value (Cn) , {e1}) ;

$$Cn := \frac{8}{\text{BesselJ} (1, \alpha_n) \alpha_n^3}$$

Таким образом, мы получили

$$C_n = \frac{8}{\alpha_n^3 J_1(\alpha_n)}.$$

Для числовых расчетов удобно образовать соответствующую функцию целочисленного аргумента:

```
> C:=unapply(Cn,n);
```

$$C := n \sim \rightarrow \frac{8}{\text{BesselJ}(1, \alpha_{n\sim}) \alpha_{n\sim}^3}$$

Искомое разложение имеет вид

$$1 - \frac{x^2}{b^2} = 8 \sum_{n=1}^{\infty} \frac{J_0\left(\frac{\alpha_n x}{b}\right)}{\alpha_n^3 J_1(\alpha_n)}.$$

Проверим теперь на численных примерах полученный результат. Сформируем отрезок ряда нашего разложения

```
> F:=(x,N,alpha)->sum(C(n)*Yn(x,n),n=1..N);
```

$$F := (x, N, \alpha) \rightarrow \sum_{n=1}^N C(n) Y_n(x, n)$$

Найдем первые несколько корней характеристического уравнения (корней функции Бесселя нулевого порядка):

```
> N:=10:alpha:=array(1..N);
```

```
> s:=BesselJZeros(0,1..N):evalf(s);
```

```
> seq(alpha[i]=%[i],i=1..N);assign(%);
```

$\alpha_1 = 2.404825558$, $\alpha_2 = 5.520078110$, $\alpha_3 = 8.653727913$, $\alpha_4 = 11.79153444$,

$\alpha_5 = 14.93091771$, $\alpha_6 = 18.07106397$, $\alpha_7 = 21.21163663$, $\alpha_8 = 24.35247153$,

$\alpha_9 = 27.49347913$, $\alpha_{10} = 30.63460647$

Здесь мы воспользовались специальной функцией `BesselJZeros()` для отыскания действительных корней функции Бесселя первого рода. Первый параметр этой функции указывает порядок (индекс) функции Бесселя, второй параметр в нашем случае указывает на необходимость найти первые десять последовательных корней функции Бесселя.

Зададим интервал разложения, например,


```
> b:=3;
```

$$b := 3$$

Построим теперь графики функций

```
> p1:=plot(f(x),x=0..b,color=black,legend=`функция f(x)`):
```

```
p2:=plot(F(x,N,alpha),x=0..b,color=black,style=point,
legend=cat(`сумма при N=`,convert(N,string))):
plots[display]({p1,p2});
```



Рассмотрим еще одну поучительную задачу.

Задача 2.20. Разложить функцию $f(x) = 1$ на отрезке $[a, b]$, $b > a > 0$ в ряд по собственным функциям задачи Штурма-Лиувилля:

$$\frac{d}{dx} \left(x \frac{dy}{dx} \right) + \lambda \frac{y}{x} = 0, \quad y(a) = 0, \quad y(b) = 0.$$

Решение. Определяем уравнение в Maple:

```
> eq:=diff(x*diff(y(x),x),x)+lambda*y(x)/x=0;
```

$$eq := \left(\frac{d}{dx} y(x) \right) + x \left(\frac{d^2}{dx^2} y(x) \right) + \frac{\lambda y(x)}{x} = 0$$

Находим общее решение уравнения

```
> res:=dsolve(eq,y(x));
```

$$res := y(x) = _C1 \sin(\sqrt{\lambda} \ln(x)) + _C2 \cos(\sqrt{\lambda} \ln(x))$$

Удобно переписать это решение в форме

```
> y:=x->
```

$$_C1 * \sin(\lambda^{1/2} * \ln(x/a)) + _C2 * \cos(\lambda^{1/2} * \ln(x/a));$$

```
simplify(eq);
```

$$y := x \rightarrow _C1 \sin\left(\sqrt{\lambda} \ln\left(\frac{x}{a}\right)\right) + _C2 \cos\left(\sqrt{\lambda} \ln\left(\frac{x}{a}\right)\right)$$

$$0 = 0$$

Здесь мы сразу же и проверили правильность представления общего решения в такой форме. Сформируем теперь уравнения по граничным условиям

```
> eq1:=y(a)=0;eq2:=y(b)=0;
```

$$eq1 := _C2 = 0$$

$$eq2 := _C1 \sin\left(\sqrt{\lambda} \ln\left(\frac{b}{a}\right)\right) + _C2 \cos\left(\sqrt{\lambda} \ln\left(\frac{b}{a}\right)\right) = 0$$

Одна константа сразу же определилась (в этом преимущество выбранной формы общего решения). А тогда для определения собственных чисел мы имеем уравнение

> Delta:=subs(_C2=0, eq2);

$$\Delta := _C1 \sin\left(\sqrt{\lambda} \ln\left(\frac{b}{a}\right)\right) = 0$$

Решаем полученное уравнение и находим собственные числа

> assume(a>0):assume(a<b):

_EnvAllSolutions:=true:

lambda:=solve(Delta, lambda);

$$\lambda := \frac{\pi^2 _Z1^2}{\ln\left(\frac{b}{a}\right)^2}$$

> lambda:=subs(_Z1='k', lambda);

$$\lambda := \frac{\pi^2 k^2}{\ln\left(\frac{b}{a}\right)^2}$$

> assume(k, posint):ev:=unapply(lambda, k);

$$ev := k \rightarrow \frac{\pi^2 k^2}{\ln\left(\frac{b}{a}\right)^2}$$

Собственные функции будут

> ef:=(x, k)->sin(Pi*k/ln(b/a)*ln(x/a));

$$ef := (x, k) \rightarrow \sin\left(\frac{\pi k \ln\left(\frac{x}{a}\right)}{\ln\left(\frac{b}{a}\right)}\right)$$

Итак, собственные числа задачи образуют дискретный спектр

$$\lambda_k = \frac{\pi^2 k^2}{\ln^2\left(\frac{b}{a}\right)}, k = 1, 2, \dots$$

и мы имеем счетное множество собственных функций

$$y_k(x) = \sin\left(\frac{\pi k \ln\left(\frac{x}{a}\right)}{\ln\left(\frac{b}{a}\right)}\right), \quad k = 1, 2, \dots$$

Проверим полученное решение. Проверим уравнение

```
> y:='y'; subs(lambda=ev(k), y(x)=ef(x,k), eq);
      y:=y
```

$$\left(\frac{\partial}{\partial x} \sin\left(\frac{\pi k \ln\left(\frac{x}{a}\right)}{\ln\left(\frac{b}{a}\right)}\right)\right) + x \left(\frac{\partial^2}{\partial x^2} \sin\left(\frac{\pi k \ln\left(\frac{x}{a}\right)}{\ln\left(\frac{b}{a}\right)}\right)\right) + \frac{\pi^2 k^2 \sin\left(\frac{\pi k \ln\left(\frac{x}{a}\right)}{\ln\left(\frac{b}{a}\right)}\right)}{\ln\left(\frac{b}{a}\right)^2 x} = 0$$

```
> simplify(value(lhs(%)));
```

0

Проверим граничные условия

```
> simplify(ef(a,k)); simplify(ef(b,k));
```

0

0

Проверим ортогональность собственных функций на отрезке $[a, b]$ с весом $\frac{1}{x}$

```
> assume(n, posint): assume(m, posint):
```

```
int(ef(x,n)*ef(x,m)/x, x=a..b); intJ:=simplify(value(%));
```

$$\int_a^b \frac{\sin\left(\frac{\pi n \ln\left(\frac{x}{a}\right)}{\ln\left(\frac{b}{a}\right)}\right) \sin\left(\frac{\pi m \ln\left(\frac{x}{a}\right)}{\ln\left(\frac{b}{a}\right)}\right)}{x} dx$$

intJ := 0

Для разложения функции нам понадобится норма. Вычислим квадрат нормы собственных функций

```
> Norma:=int(ef(x,n)^2/x, x=a..b);
```

```
Norma:=simplify(value(Norma));
```

$$Norma := \int_{a\sim}^{b\sim} \frac{\sin\left(\frac{\pi n\sim \ln\left(\frac{x}{a\sim}\right)}{\ln\left(\frac{b\sim}{a\sim}\right)}\right)^2}{x} dx$$

$$Norma := \frac{1}{2} \ln\left(\frac{b\sim}{a\sim}\right)$$

Итак,

$$\|y_n\|^2 = \frac{1}{2} \ln\left(\frac{b}{a}\right).$$

Приступим теперь к разложению заданной функции

$$> f := x \rightarrow 1;$$

$$f := 1$$

Коэффициенты разложения, очевидно, находятся по формуле

$$C_n = \frac{1}{\|y_n\|^2} \int_a^b \frac{1}{x} f(x) y_n(x) dx.$$

Таким образом,

$$> Cn := \text{Int}(f(x) * e f(x, n) / x, x=a..b) / Norma;$$

$$Cn := \frac{2}{\ln\left(\frac{b\sim}{a\sim}\right)} \int_{a\sim}^{b\sim} \frac{\sin\left(\frac{\pi n\sim \ln\left(\frac{x}{a\sim}\right)}{\ln\left(\frac{b\sim}{a\sim}\right)}\right)}{x} dx$$

$$> Cn := \text{simplify}(\text{value}(Cn));$$

$$Cn := -\frac{2((-1)^{n\sim} - 1)}{\pi n\sim}$$

Итак, мы нашли

$$C_n = \frac{2[1 - (-1)^n]}{\pi n}.$$

Следовательно, искомое разложение имеет вид

$$1 = \sum_{n=1}^{\infty} \frac{2[1 - (-1)^n]}{\pi n} \sin\left(\frac{\pi n \ln\left(\frac{x}{a}\right)}{\ln\left(\frac{b}{a}\right)}\right).$$

Проверим теперь на численных примерах полученный результат.

Удобно сформировать функцию целочисленного аргумента

```
> C:=unapply(Cn, n);
```

$$C := n \sim \rightarrow -\frac{2((-1)^{n\sim} - 1)}{\pi n\sim}$$

Образует отрезок нашего ряда в виде функции

```
> F:=(x, N) -> sum(C(n) * ef(x, n), n=1..N);
```

$$F := (x, N) \rightarrow \sum_{n=1}^N C(n) \text{ef}(x, n)$$

Зададим интервал разложения, например.

```
> a:=1;b:=2;
```

```
a := 1
```

```
b := 2
```

Построим графики функций


```
> p:=plot(f(x), x=a..b, color=black, legend='функция f(x)');
```

```
N:=10;
```

```
p10:=plot(F(x, N), x=a..b, color=black, linestyle=3,
```

```
legend=cat('сумма при N=', convert(N, string))):
```

```
plots[display]({p, p10});
```




Увеличим число членов разложения

```

> N:=20:
p20:=plot(F(x,N),x=a..b,color=black,style=point,
symbol=cross,legend=cat(`сумма при N=`,convert(N,string))):
N:=40:
p40:=plot(F(x,N),x=a..b,color=black,
style=point,symbol=circle,
legend=cat(`сумма при N=`,convert(N,string))):
plots[display]({p,p10,p20,p40});

```



На графиках мы видим отчетливое проявление особенности поведения частных сумм, так называемое явление Гиббса. Вследствие хорошо известной теоремы из теории функций, ряд Фурье не может равномерно сходиться в окрестности точки разрыва. Влияние разрывности состоит в том, что возникают колебания вокруг истинного значения, и амплитуды этих колебаний убывают очень медленно с увеличением числа членов. Колебания высокой частоты усеченного ряда вокруг истинного значения функции всегда имеют место. Обычно амплитуды этих колебаний достаточно малы, чтобы не вызывать особых последствий. Однако в случае разрыва они весьма заметны и мешают эффективному гармоническому синтезу. Чтобы сгладить эти колебания можно использовать так называемые σ -множители Ланцоша, которые добавляются к каждому слагаемому частичной суммы. Эти множители имеют вид

$$\sigma_k = \frac{\sin\left(\frac{\pi k}{N}\right)}{\frac{\pi k}{N}},$$

где N — количество членов суммы. Посмотрим, как сглаживаются нежелательные колебания в нашем случае. Для этого образуем частичные суммы по Ланцошу


```
> Fs := (x, N) ->
```

```
sum(C(n) * (sin(n*Pi/N) / (n*Pi/N)) * ef(x, n), n=1..N);
```

$$Fs := (x, N) \rightarrow \sum_{n=1}^N \frac{C(n) \sin\left(\frac{n\pi}{N}\right) N \text{ef}(x, n)}{n\pi}$$


Теперь посмотрим на графиках результат такого сглаживания:

```
> N:=20:
```

```
ps:=plot(Fs(x, N), x=a..b, color=black, thickness=3,
```

```
legend=cat(`сумма Ланцоша при N=`, convert(N, string))):
```

```
plots[display]({p, ps, p20, p40});
```




```
> N:=40:
```

```
ps1:=plot(Fs(x, N), x=a..b, color=black, thickness=3,
```

```
legend=cat(`сумма Ланцоша при N=`, convert(N, string))):
```

```
plots[display]({p, ps1, p40});
```



Улучшить сходимость можно и с помощью множителей Фейера $\left(1 - \frac{n}{N}\right)$. Такой подход полностью устраняет колебания. Приближение к функции, полученное по этому методу, осуществляется с помощью совершенно гладких монотонных функций, которые остаются постоянно ниже первоначальной кривой. Посмотрим на этот способ в нашем случае:

```
> Ff := (x, N) -> sum((1 - n/N) * C(n) * ef(x, n), n = 1..N);
```


$$Ff := (x, N) \rightarrow \sum_{n=1}^N \left(1 - \frac{n}{N}\right) C(n) \text{ef}(x, n)$$

```
> N := 20;
```

```
pf1 := plot(Ff(x, N), x = a..b, color = black, thickness = 3,
  legend = cat(`сумма Фейера при N = `, convert(N, string)),
  linestyle = 3);
```

```
N := 40;
```

```
pf2 := plot(Ff(x, N), x = a..b, color = black, thickness = 3,
  legend = cat(`сумма Фейера при N = `, convert(N, string)));
> plots[display]({p, pf1, pf2, p40});
```



Упражнения

1. Дано уравнение $y'' + y = 0$, решить следующие краевые задачи:

1.1. $y(0) = 1, y\left(\frac{\pi}{2}\right) = 0$;

1.2. $y(0) = 0, y\left(\frac{\pi}{2}\right) = 0$;

1.3. $y(0) = 1, y(\pi) = 2$;

1.4. $y(0) = 0, y(\pi) = 0$.

Ответ. 1.1. $y = \cos x$; 1.2. $y = 0$; 1.3. задача не разрешима; 1.4. $y = C \sin x$.

2. При каких значениях параметра λ разрешима однородная краевая задача $y'' + \lambda y = 0, y(0) = 0, y(\pi) = 0$?

Ответ. Задача разрешима для бесконечной последовательности значений $\lambda_n = n^2, n = 1, 2, \dots$; решениями задачи являются функции $y_n = C \sin nx$.

3. Найти решения уравнения Бесселя нулевого порядка $xy'' + y' + xy = 0$ при краевых условиях $y(0)$ – ограничено, $y(l) = 1$.

Ответ. $y(x) = \frac{J_0(x)}{J_0(l)}$, если только $J_0(l) \neq 0$.

4. Найти решение уравнения $y'' + 2y' + y = e^x$, удовлетворяющее краевым условиям $y(0) = 0, y'(2) = 0$.

Ответ. $y = \frac{e^x - e^{-x}}{4} + \frac{(1 + e^4)xe^{-x}}{4}$.

5. Разрешима ли однородная краевая задача $y'' + 2y' + y = 0, y(0) = 0, y'(2) = 0$? Какая имеется связь между данной краевой задачей и задачей 4?

Ответ. Нет.

6. Определить, для каких значений l разрешима однородная краевая задача $xy'' + y' + xy = 0, y(0)$ – ограничено, $y(l) = 0$. (Сравните с решением задачи 3).

Ответ. l — корень уравнения $J_0(x) = 0$.

7. Определить, для каких значений λ разрешима однородная краевая задача $y'' + \lambda y = 0$, $y(0) = 0$, $y(b) = 0$.

Ответ. $\lambda_n = \left(\frac{n\pi}{b}\right)^2$, $n = 1, 2, \dots$

8. Найти систему собственных функций краевой задачи $y'' + \lambda y = 0$, $y(0) = y(2\pi)$, $y'(0) = y'(2\pi)$.

Ответ. $\lambda_n = n^2$, $y_n = a_n \cos nx + b_n \sin nx$, $n = 0, 1, 2, \dots$

9. Найти систему собственных функций однородной краевой задачи $y'' + \frac{1}{x}y' + \left(\lambda - \frac{\nu^2}{x^2}\right)y = 0$, $y(0)$ – ограничено, $y(l) = 0$.

Ответ. $y_n(x) = J_\nu\left(\frac{\alpha_n x}{l}\right)$, $n = 1, 2, \dots$, где α_n — n -ый положительный нуль функции $J_\nu(x)$.

10. Материальная точка массы m движется по оси Ox , находясь под действием силы $(-bx)$ притягивающей ее к началу координат, силы сопротивления среды $\left(-a \frac{dx}{dt}\right)$ и внешней силы, направленной по оси Ox и равной $F(t)$ в момент времени t . Составить дифференциальное уравнение движения этой точки.

Указание. Воспользоваться законом Ньютона, согласно которому произведение массы тела на ускорение равно силе, действующей на тело. Ввести обозначения: $h = \frac{a}{2m}$, $k^2 = \frac{b}{m}$, $f(t) = \frac{F(t)}{m}$.

Ответ. $x'' + 2hx' + k^2x = f(t)$.

11. Проинтегрировать дифференциальное уравнение предыдущей задачи 10 в предположении, что $f(t) = 0$, $h = 0$ (случай свободных колебаний в среде без сопротивления). Найти период и частоту колебаний. Найти амплитуду и начальную фазу колебания, если заданы начальные условия: $x = x_0$, $x' = v_0$ при $t = 0$. Рассмотреть случай $k = 1$, $x_0 = 0$, $v_0 = 0$.

Ответ. $x = A \sin(kt + \varphi)$, k — частота колебаний, $T = \frac{2\pi}{k}$ — период, A — амплитуда, φ — начальная фаза; $A = \sqrt{x_0^2 + \frac{v_0^2}{k^2}}$, $\varphi = \arctg \frac{kx_0}{v_0}$, $x = \sin t$.

12. Проинтегрировать дифференциальное уравнение задачи 10 в предположении, что $f(t) = 0$, $h > 0$ (случай свободных колебаний в среде с сопротивлением, пропорциональным первой степени скорости). В каком случае (при каком соотношении между k и h) движение точки будет затухающим гармоническим колебанием? Найти начальную амплитуду (значение амплитуды при $t = 0$) и начальную фазу колебания, если заданы начальные условия: $x = 1$, $x' = 2$ при $t = 0$, считая, что $k = 2$, $h = 1$.

Ответ. $x = C_1 e^{(-h + \sqrt{h^2 - k^2})t} + C_2 e^{(-h - \sqrt{h^2 - k^2})t}$, при $h^2 - k^2 > 0$; $x = e^{-ht} (C_1 + C_2 t)$, при $h^2 = k^2$; $x = A e^{-ht} \sin(\sqrt{k^2 - h^2} t + \varphi)$, при $h^2 - k^2 < 0$; $\sqrt{k^2 - h^2}$ — частота колебаний, $T = \frac{2\pi}{\sqrt{k^2 - h^2}}$ — период, $A e^{-ht}$ — амплитуда, A — начальная амплитуда, $A = 2$, $\varphi = \frac{\pi}{6}$.

13. Проинтегрировать дифференциальное уравнение задачи 10 в предположении, что $f(t) = M \sin \omega t$, $h = 0$ (случай вынужденных колебаний в среде без сопротивления при наличии синусоидальной внешней силы). Рассмотреть случаи $\omega \neq k$ и $\omega = k$ (резонанс).

Ответ. $x = \frac{M}{k^2 - \omega^2} \sin \omega t + A \sin(kt + \varphi)$ при $\omega \neq k$;
 $x = -\frac{M}{2k} t \cos kt + A \sin(kt + \varphi)$ при $\omega = k$.

14. Доказать, что решением уравнения $x'' + k^2 x = f(t)$, с нулевыми начальными значениями искомой функции и ее производной при $t = 0$ будет

$$x = \frac{1}{k} \int_0^t f(\xi) \sin k(t - \xi) d\xi.$$

15. При каком условии относительно ω общее решение уравнения $x'' + x = \cos \omega t$ не будет иметь векового¹ члена?

Ответ. $\omega \neq 1$.

16. При каких значениях k общее решение уравнения $x'' + k^2 x = \cos 2t$ не имеет векового члена?

Ответ. $k \neq \sqrt{2}$.

¹ Вековым членом называется член, имеющий вид произведения периодической функции на степень независимой переменной.

17. При каких значениях h все ненулевые решения уравнения $x'' + 2hx' + x = 0$ представляют собой затухающие гармонические колебания?

Ответ. $0 < h < 1$.

18. При каких значениях k все ненулевые решения уравнения $x'' + 2x' + k^2x = 0$ представляют собой затухающие гармонические колебания?

Ответ. $k > 1$.

19. При каких значениях q уравнение $y'' + qy = 0$ имеет ненулевые решения, стремящиеся к нулю при $x \rightarrow +\infty$?

Ответ. $q < 0$.

20. При каких значениях p и q все решения уравнения $y'' + py' + qy = 0$ ограничены при всех значениях $x \geq 0$?

Ответ. $p \geq 0, q > 0$.

21. При каких значениях p и q все решения уравнения $y'' + py' + qy = 0$ стремятся к нулю при $x \rightarrow +\infty$?

Ответ. $p > 0, q > 0$.

22. При каких значениях p и q все решения уравнения $y'' + py' + qy = 0$ являются периодическими функциями от x ?

Ответ. $p = 0, q > 0$.

23. Построить фундаментальную систему решений заданного уравнения, нормированную в точке $x = 0$: а) $y'' + k^2y = 0$; б) $y'' - k^2y = 0$.

Указание. Фундаментальная система решений $y_1(x)$ и $y_2(x)$, нормированная в точке $x = x_0$, удовлетворяет граничным условиям $y_1(x_0) = 1, y_1'(x_0) = 0$; $y_2(x_0) = 0, y_2'(x_0) = 1$.

Ответ. а) $y_1 = \cos kx, y_2 = \frac{\sin kx}{k}$; б) $y_1 = \operatorname{ch} kx, y_2 = \frac{\operatorname{sh} kx}{k}$.

24. Найти собственные значения и собственные функции краевой задачи

24.1. $y'' + \lambda y = 0, y(0) = 0, y(l) = 0, x \in [0, l]$;

24.2. $y'' + \lambda y = 0, y(0) = 0, y'(l) + hy(l) = 0, x \in [0, l], h > 0$;

24.3. $y'' + \lambda y = 0, y(l) = 0, -y'(0) + hy(0) = 0, x \in [0, l], h > 0$;

24.4. $y'' + \lambda y = 0, y(0) = 0, y'(l) = 0, x \in [0, l]$;

$$24.5. y'' + \lambda y = 0, y'(0) = 0, y(l) = 0, x \in [0, l];$$

$$24.6. y'' + \lambda y = 0, y'(0) = 0, y'(l) = 0, x \in [0, l];$$

$$24.7. y'' + \lambda y = 0, y'(0) = 0, y'(l) + hy(l) = 0, x \in [0, l], h > 0;$$

$$24.8. y'' + \lambda y = 0, y'(l) = 0, -y'(0) + hy(0) = 0, x \in [0, l], h > 0;$$

$$24.9. y'' + \lambda y = 0, -y'(0) + hy(0) = 0, y'(l) + hy(l) = 0, x \in [0, l], h > 0;$$

$$24.10. \frac{d}{dx} \left(x \frac{dy}{dx} \right) + \frac{\lambda}{x} y = 0, y(a) = 0, y(b) = 0, x \in [a, b], a \neq 0, b \neq 0;$$

$$24.11. \frac{d}{dx} \left(x^2 \frac{dy}{dx} \right) + \lambda x^2 y = 0, y(a) = 0, y(b) = 0, x \in [a, b], a \neq 0, b \neq 0;$$

$$24.12. y'' + \lambda y = 0, y(-l) = 0, y(l) = 0, x \in [-l, l];$$

$$24.13. y'' + \lambda y = 0, y'(-l) = 0, y'(l) = 0, x \in [-l, l];$$

$$24.14. y'' + \lambda y = 0, -y'(-l) + hy(-l) = 0, y'(l) + hy(l) = 0, x \in [-l, l], h > 0;$$

$$24.15. \frac{d}{dx} \left(x \frac{dy}{dx} \right) + \frac{\lambda}{x} y = 0, y'(a) = 0, y'(b) = 0, x \in [a, b], a \neq 0, b \neq 0.$$

25. Разложить функцию $f(x) = x^2$ в ряд по собственным функциям задачи

$$y'' + \lambda y = 0, y(-l) = y(l), y'(-l) = y'(l), x \in [-l, l].$$

26. Разложить функцию

$$f(x) = \begin{cases} 0, & -l \leq x \leq 0, \\ x^2, & 0 \leq x \leq l \end{cases}$$

в ряд по собственным функциям задачи

$$y'' + \lambda y = 0, y(-l) = y(l), y'(-l) = y'(l), x \in [-l, l];$$

27. Разложить функцию $f(x) = x^2$ в ряд по собственным функциям задачи

$$y'' + \lambda y = 0, y(0) = 0, y(l) = 0, x \in [0, l].$$

28. Разложить функцию $f(x) = 1$ в ряд по собственным функциям задачи 24.10.

29. Разложить функцию $f(x) = x^2$ в ряд по собственным функциям задачи

$$\frac{d}{dx} \left(x^2 \frac{dy}{dx} \right) + \lambda x^2 y = 0, y(0) = O(1), y'(a) = 0, x \in [0, a].$$

30. Разложить функцию $f(x) = c^2 - x^2$ в ряд по собственным функциям задачи

$$\frac{d}{dx} \left(x^2 \frac{dy}{dx} \right) + \lambda x^2 y = 0, \quad y(0) = O(1), \quad y'(a) + hy(a) = 0, \quad x \in [0, a], \quad h > 0.$$

31. Разложить функцию $f(x) = 1 + cx$ в ряд по собственным функциям задачи

$$y'' + \lambda y = 0, \quad -y'(0) + hy(0) = 0, \quad y(a) = 0, \quad x \in [0, a], \quad h > 0.$$

32. Разложить функцию $f(x) = x$ в ряд по собственным функциям задачи 24.10.

33. Решить сингулярные задачи Штурма–Лиувилля для уравнения

$$\frac{d}{dx} \left(x^2 \frac{dy}{dx} \right) + \lambda x^2 y = 0, \quad x \in (0, a), \quad h > 0$$

при условиях

$$33.1. \quad y|_{x=a} = 0, \quad y|_{x=0} = O(1);$$

$$33.2. \quad y'|_{x=a} = 0, \quad y|_{x=0} = O(1);$$

$$33.3. \quad y' + hy|_{x=a} = 0, \quad y|_{x=0} = O(1), \quad h > 0.$$

34. Найти функции Грина следующих регулярных дифференциальных операторов второго порядка

$$34.1. \quad -y'' = 0, \quad y(0) = 0, \quad y(1) = 0, \quad x \in [0, 1];$$

$$34.2. \quad -y'' = 0, \quad y(0) = 0, \quad y'(1) = 0, \quad x \in [0, 1];$$

$$34.3. \quad -y'' = 0, \quad y(0) = 0, \quad y'(1) + hy(1) = 0, \quad x \in [0, 1];$$

$$34.4. \quad -y'' - y = 0, \quad y(0) = 0, \quad y(1) = 0, \quad x \in [0, 1];$$

$$34.5. \quad -y'' + y = 0, \quad y(0) = 0, \quad y(1) = 0, \quad x \in [0, 1];$$

$$34.6. \quad -y'' + y = 0, \quad y'(0) = 0, \quad y'(1) = 0, \quad x \in [0, 1];$$

$$34.7. \quad -y'' + (1 + x^2)y = 0, \quad y(0) = 0, \quad y'(1) = 0, \quad x \in [0, 1];$$

$$34.8. \quad -y'' - y = 0, \quad -y'(0) + y(0) = 0, \quad -y'(1) + y(1) = 0, \quad x \in [0, 1];$$

$$34.9. \quad -\left(\frac{1}{x-2} y' \right)' + \frac{3y}{(x-2)^3} = 0, \quad y(0) = 0, \quad y(1) = 0, \quad x \in [0, 1];$$

$$34.10. \quad -(1 + x^2)y'' - 2xy' = 0, \quad y(0) = 0, \quad y'(1) + y(1) = 0, \quad x \in [0, 1];$$

$$\text{Ответ. 34.1. } G(x, y) = \begin{cases} x(1-y), & x \leq y \\ y(1-x), & x \geq y \end{cases}; \quad 34.2. \quad G(x, y) = \begin{cases} x, & x \leq y \\ y, & x \geq y \end{cases};$$

$$34.3. \quad G(x, y) = \begin{cases} \frac{x(1+h-y)}{1+h}, & x \leq y \\ \frac{y(1+h-x)}{1+h}, & x \geq y \end{cases};$$

$$34.4. \quad G(x, y) = \begin{cases} \frac{\sin x (\cos y \sin 1 - \sin y \cos 1)}{\sin 1}, & x \leq y \\ \frac{\sin y (\cos x \sin 1 - \sin x \cos 1)}{\sin 1}, & x \geq y \end{cases};$$

$$34.5. \quad G(x, y) = \begin{cases} \frac{(e^x - e^{-x})(e^y - e^{(2-y)})}{2(1-e^2)}, & x \leq y \\ \frac{(e^y - e^{-y})(e^x - e^{(2-x)})}{2(1-e^2)}, & x \geq y \end{cases};$$

$$34.6. \quad G(x, y) = \begin{cases} \frac{(e^x + e^{-x})(e^y + e^{(2-y)})}{2(1-e^2)}, & x \leq y \\ \frac{(e^y + e^{-y})(e^x + e^{(2-x)})}{2(1-e^2)}, & x \geq y \end{cases};$$

$$34.7. \quad G(x, y) = \begin{cases} \frac{e^{-\frac{x^2+y^2}{2}} \Phi(x) (\Phi(y)\sqrt{\pi} - \Phi(1)\sqrt{\pi} - 2e^{-1})\sqrt{\pi}}{\Phi(1)\sqrt{\pi} + 2e^{-1}}, & x \leq y \\ \frac{e^{-\frac{x^2+y^2}{2}} \Phi(y) (\Phi(x)\sqrt{\pi} - \Phi(1)\sqrt{\pi} - 2e^{-1})\sqrt{\pi}}{\Phi(1)\sqrt{\pi} + 2e^{-1}}, & x \geq y \end{cases},$$

где $\Phi(x)$ — интеграл вероятности.

34.8.

$$G(x, y) = \begin{cases} -\frac{(\sin x + \cos x)(\sin 2 \sin y + \sin y - \cos y + 2 \cos y \cos^2 1)}{(2 - 2 \cos^2 1 + \sin 2)}, & x \leq y \\ -\frac{(\sin y + \cos y)(\sin 2 \sin x + \sin x - \cos x + 2 \cos x \cos^2 1)}{(2 - 2 \cos^2 1 + \sin 2)}, & x \geq y \end{cases};$$

$$34.9. G(x, y) = \begin{cases} \frac{x(x^3 - 8x^2 + 24x - 32)(y^4 - 8y^3 + 24y^2 - 32y + 15)}{60(x-2)(y-2)}, & x \leq y \\ \frac{y(y^3 - 8y^2 + 24y - 32)(x^4 - 8x^3 + 24x^2 - 32x + 15)}{60(y-2)(x-2)}, & x \geq y \end{cases};$$

$$34.10. G(x, y) = \begin{cases} -\frac{\operatorname{arctg} x(-\pi - 2 + 4 \operatorname{arctg} y)}{\pi + 2}, & x \leq y \\ -\frac{\operatorname{arctg} y(-\pi - 2 + 4 \operatorname{arctg} x)}{\pi + 2}, & x \geq y \end{cases}.$$

3. Классификация уравнений в частных производных второго порядка

Математическое описание физических процессов приводит к дифференциальным и интегральным уравнениям или даже к интегро-дифференциальным уравнениям. Широкий класс физических процессов описывается линейными дифференциальными уравнениями второго порядка

$$\sum_{i=1}^n \sum_{j=1}^n a_{ij}(x) \frac{\partial^2 u}{\partial x_i \partial x_j} + \sum_{i=1}^n b_i(x) \frac{\partial u}{\partial x_i} + c(x)u = F(x). \quad (3.1)$$

Прежде чем приступить к рассмотрению задач классификации таких уравнений, напомним некоторые определения [3].

Классификация уравнений в точке

Будем рассматривать функции n независимых переменных $u = u(x_1, x_2, \dots, x_n)$. Рассмотрим квазилинейное (т.е. линейное относительно старших производных) дифференциальное уравнение второго порядка

$$\sum_{i,j=1}^n a_{ij}(x) \frac{\partial^2 u}{\partial x_i \partial x_j} + \Phi(x, u, \nabla u) = 0 \quad (3.2)$$

с непрерывными коэффициентами $a_{ij}(x)$ в области D . Можно считать без потери общности, что $a_{ij} = a_{ji}$. Зафиксируем определенную точку $(x_1^0, x_2^0, \dots, x_n^0)$ в области D и составим квадратичную форму

$$g(X_1, X_2, \dots, X_n) = \sum_{i,j=1}^n a_{ij}(x^0) X_i X_j. \quad (3.3)$$

Квадратичная форма (3.3) называется квадратичной формой, связанной с уравнением (3.2). Составим определитель формы (3.3)

$$\Delta = \begin{vmatrix} a_{11} & a_{12} & \cdots & a_{1n} \\ a_{21} & a_{22} & \cdots & a_{2n} \\ \cdots & \cdots & \cdots & \cdots \\ a_{n1} & a_{n2} & \cdots & a_{nn} \end{vmatrix}.$$

Будем говорить, что квадратичная форма (3.3) определенная, если она сохраняет знак при любых изменениях переменных X и неопределенная, если ее знак меняется.

Мы скажем, что уравнение (3.2) принадлежит *эллиптическому типу* в точке $(x_1^0, x_2^0, \dots, x_n^0)$, если $\Delta \neq 0$ и g — определенная форма; уравнение (3.2) принадлежит *гиперболическому типу*, если $\Delta \neq 0$ и g — неопределенная форма; уравнение (3.2) принадлежит *параболическому типу*, если $\Delta = 0$. Уравнение (3.2) принадлежит эллиптическому типу (соответственно, гиперболическому или параболическому) в области D , если во всех точках этой области оно принадлежит эллиптическому (соответственно, гиперболическому или параболическому) типу. Если коэффициенты a_{ij} постоянные, то принадлежность уравнения к тому или иному типу не зависит от значений независимых переменных.

Отметим, что приведенная классификация не зависит от выбора переменных и остается неизменной при замене переменных, однако, она зависит от точки x^0 .

Уравнения разных типов описывают разные по характеру явления и процессы и, наоборот, уравнения одного типа описывают схожие явления. Например, гиперболические уравнения описывают процессы распространения волн и колебания; параболические уравнения описывают явления переноса; эллиптические уравнения — статические процессы, задачи об установившихся стационарных движениях.

Рассмотрим ряд примеров.

1. Уравнение Лапласа

$$\Delta u = \frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} + \frac{\partial^2 u}{\partial z^2} = 0.$$

Составим для него квадратичную форму $g = X^2 + Y^2 + Z^2$; определитель этой формы

$$\Delta = \begin{vmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{vmatrix} = 1 \neq 0.$$

Ясно, что g — определенная форма (положительно определенная). Следовательно, уравнение Лапласа принадлежит к эллиптическому типу. К этому же типу относятся уравнения Пуассона и Гельмгольца.

2. Волновое уравнение

$$\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} + \frac{\partial^2 u}{\partial z^2} - \frac{1}{a^2} \frac{\partial^2 u}{\partial t^2} = 0.$$

Составим для него квадратичную форму $g = X^2 + Y^2 + Z^2 - T^2/a^2$; определитель этой формы

$$\Delta = \begin{vmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -\frac{1}{a^2} \end{vmatrix} = -\frac{1}{a^2} \neq 0.$$

Очевидно, что g — неопределенная форма; тип уравнения — гиперболический.

3. Уравнение Фурье

$$\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} + \frac{\partial^2 u}{\partial z^2} - \frac{1}{a^2} \frac{\partial u}{\partial t} = 0.$$

Квадратичная форма $g = X^2 + Y^2 + Z^2$; определитель этой формы

$$\Delta = \begin{vmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 \end{vmatrix} = 0.$$

Уравнение относится к параболическому типу.

4. Уравнение Трикоми

$$y \frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} = 0.$$

Квадратичная форма $g = yX^2 + Y^2$; определитель этой формы

$$\Delta = \begin{vmatrix} y & 0 \\ 0 & 1 \end{vmatrix} = y.$$

Это уравнение *смешанного типа*. Если $y < 0$ — тип уравнения гиперболический; если $y > 0$ — эллиптический тип, а если $y = 0$, то уравнение параболического типа.

Классификация уравнений с двумя независимыми переменными

Для каждой точки x^0 имеется, вообще говоря, свое преобразование независимых переменных, приводящее уравнение к каноническому виду. Дифференциальное уравнение с числом независимых переменных больше двух (если исключить случай постоянных коэффициентов), вообще говоря, невозможно привести с помощью преобразования независимых переменных к каноническому виду даже в сколь угодно малой области. В случае же двух независимых переменных такое преобразование независимых переменных существует при достаточно общих предположениях о коэффициентах уравнения.

Для случаев двух, трех или четырех независимых переменных классификацию уравнений можно упростить.

Рассмотрим квазилинейное уравнение второго порядка с двумя независимыми переменными

$$A(x, y) \frac{\partial^2 u}{\partial x^2} + 2B(x, y) \frac{\partial^2 u}{\partial x \partial y} + C(x, y) \frac{\partial^2 u}{\partial y^2} + \Phi \left(x, y, u, \frac{\partial u}{\partial x}, \frac{\partial u}{\partial y} \right) = 0, \quad (3.4)$$

где коэффициенты A , B и C суть функции от x и y , имеющие непрерывные производные до второго порядка включительно. Будем предполагать, что A , B и C не обращаются одновременно в нуль. Уравнению (3.4) соответствует квадратичная форма

$$g = AX^2 + 2BXY + CY^2,$$

определитель которой есть

$$\Delta = \begin{vmatrix} A & B \\ B & C \end{vmatrix} = AC - B^2.$$

Пусть $\Delta = AC - B^2 > 0 \Rightarrow A \neq 0$. Тогда можно записать

$$g = \frac{1}{A} \left\{ (AX + BY)^2 + (AC - B^2)Y^2 \right\}. \quad (3.5)$$

Форма (3.5), очевидно, определенная. Следовательно, уравнение принадлежит эллиптическому типу.

Пусть $\Delta = AC - B^2 < 0$. Возможны два случая:

1. $A \neq 0$, тогда имеет место равенство (3.5), но в зависимости от Y знак g меняется; форма неопределенная. Следовательно, уравнение (3.4) — гиперболического типа.
2. $A = 0$, тогда запишем g в виде $g = (2BX + CY)Y$. Ясно, что g — неопределенная форма, тип уравнения — гиперболический.

Пусть $\Delta = AC - B^2 = 0$. Тип уравнения — параболический.

Таким образом, окончательно получаем: если $\Delta > 0$, то уравнение эллиптического типа; если $\Delta < 0$, то уравнение гиперболического типа, если $\Delta = 0$, то уравнение параболического типа.

Преобразование уравнений второго порядка с помощью замены переменных

Всякое линейное дифференциальное уравнение второго порядка с двумя независимыми переменными может быть записано в виде

$$A \frac{\partial^2 u}{\partial x^2} + 2B \frac{\partial^2 u}{\partial x \partial y} + C \frac{\partial^2 u}{\partial y^2} + D \frac{\partial u}{\partial x} + E \frac{\partial u}{\partial y} + Fu + G = 0, \quad (3.6)$$

где A, B, C, D, E, F, G — заданные функции от x и y (или, в частном случае, постоянные).

Уравнение (3.6) можно упростить с помощью замены переменных

$$\begin{cases} \xi = \varphi(x, y); \\ \eta = \psi(x, y). \end{cases} \quad (3.7)$$

Здесь ξ и η — новые независимые переменные. Будем считать, что отображение (3.7) является взаимно однозначным, и сделаем требуемую замену переменных. Мы получим снова *линейное уравнение второго порядка* с неизвестной функцией u и независимыми переменными ξ и η .

$$\bar{A} \frac{\partial^2 u}{\partial \xi^2} + 2\bar{B} \frac{\partial^2 u}{\partial \xi \partial \eta} + \bar{C} \frac{\partial^2 u}{\partial \eta^2} + \Phi \left(\xi, \eta, u, \frac{\partial u}{\partial \xi}, \frac{\partial u}{\partial \eta} \right) = 0, \quad (3.8)$$

где

$$\begin{aligned} \bar{A} &= A \left(\frac{\partial \xi}{\partial x} \right)^2 + 2B \frac{\partial \xi}{\partial x} \frac{\partial \xi}{\partial y} + C \left(\frac{\partial \xi}{\partial y} \right)^2; \\ \bar{B} &= A \frac{\partial \xi}{\partial x} \frac{\partial \eta}{\partial x} + B \left(\frac{\partial \xi}{\partial x} \frac{\partial \eta}{\partial y} + \frac{\partial \eta}{\partial x} \frac{\partial \xi}{\partial y} \right) + C \frac{\partial \xi}{\partial y} \frac{\partial \eta}{\partial y}; \\ \bar{C} &= A \left(\frac{\partial \eta}{\partial x} \right)^2 + 2B \frac{\partial \eta}{\partial x} \frac{\partial \eta}{\partial y} + C \left(\frac{\partial \eta}{\partial y} \right)^2; \end{aligned}$$

а функция Φ линейна относительно u, u'_ξ, u'_η .

Уравнение (3.8) становится особенно простым, если в нем коэффициенты \bar{A} и \bar{C} окажутся равными нулю. Для того чтобы первоначально заданное уравнение (3.6) можно было привести к такому простому виду, надо в нем сделать замену переменных (3.7), подобрав функции φ и ψ так, чтобы они являлись решениями уравнения

$$A \left(\frac{\partial z}{\partial x} \right)^2 + 2B \frac{\partial z}{\partial x} \frac{\partial z}{\partial y} + C \left(\frac{\partial z}{\partial y} \right)^2 = 0. \quad (3.9)$$

Это уравнение является нелинейным уравнением в частных производных первого порядка. Следующая теорема показывает, как связаны его решения с общим решением некоторого обыкновенного уравнения.

Теорема. Для того чтобы функция $z = f(x, y)$ во всех точках области Ω удовлетворяла уравнению (3.9), необходимо и достаточно, чтобы семейство $f(x, y) = \text{const}$ было общим интегралом уравнения

$$A(dy)^2 - 2Bdx dy + C(dx)^2 = 0 \quad (3.10)$$

в той же области Ω .

Доказательство этой теоремы можно найти, например, в книге Ю.С. Очана [7]. Теорема открывает путь для упрощения исходного уравнения (3.6). Для этого сначала составляем вспомогательное уравнение (3.10); оно называется *характеристическим уравнением* для данного уравнения (3.6). Характеристическое уравнение есть *обыкновенное* дифференциальное уравнение первого порядка, но второй степени. Разрешая его относительно производной y' , получим два уравнения

$$y' = \frac{B + \sqrt{B^2 - AC}}{A}; \quad (3.11)$$

$$y' = \frac{B - \sqrt{B^2 - AC}}{A}; \quad (3.12)$$

Если общий интеграл уравнения (3.11) имеет вид $\varphi(x, y) = const$, то, полагая $\xi = \varphi(x, y)$, мы обращаем в нуль коэффициент при производной $u''_{\xi\xi}$. Если $\psi(x, y) = const$ является общим интегралом уравнения (3.12), независимым от интеграла $\varphi(x, y) = const$, то, полагая $\eta = \psi(x, y)$, мы обратим в нуль также и коэффициент при производной $u''_{\eta\eta}$.

Интегральные кривые характеристического уравнения, то есть все кривые, входящие в семейства $\varphi(x, y) = const$ и $\psi(x, y) = const$, называются *характеристиками* заданного дифференциального уравнения (3.6). В связи с этим рассматриваемый метод упрощения уравнения (3.6) называется *методом характеристик*.

Уравнение гиперболического типа

Семейства $\varphi(x, y) = const$ и $\psi(x, y) = const$ можно рассматривать, как общие интегралы уравнения (3.10) — это уравнение распадается на два уравнения (3.11) и (3.12). Правые части уравнений (3.11) и (3.12) действительны и различны. Следовательно, согласно указанной теореме, функции $z = \varphi(x, y)$ и $z = \psi(x, y)$ являются решениями уравнения в частных производных (3.9). Функции $\varphi(x, y)$ и $\psi(x, y)$ линейно независимы (можно доказать, что их определитель Вронского отличен от нуля, если $AC - B^2 < 0$). Поэтому, возвращаясь к уравнению (3.6), мы можем в нем сделать замену переменных по формулам (3.7). Так как функции $\varphi(x, y)$ и $\psi(x, y)$ удовлетворяют уравнению (3.10), то в результате этой замены переменных окажется $\bar{A} = 0$ и $\bar{C} = 0$. Следовательно, уравнение (3.6) преобразуется к виду

$$2\bar{B} \frac{\partial^2 u}{\partial \xi \partial \eta} + \Phi \left(\xi, \eta, u, \frac{\partial u}{\partial \xi}, \frac{\partial u}{\partial \eta} \right) = 0,$$

или, после деления на $2\bar{B}$ и переноса в другую часть равенства, к виду

$$\frac{\partial^2 u}{\partial \xi \partial \eta} = \bar{\Phi} \left(\xi, \eta, u, \frac{\partial u}{\partial \xi}, \frac{\partial u}{\partial \eta} \right). \quad (3.13)$$

Полученное уравнение имеет более простой вид, чем исходное уравнение (3.6); если мы его сможем решить, то для того, чтобы найти решение исходного уравнения, достаточно вернуться к старым переменным.

Уравнение (3.13) представляет каноническую форму уравнения гиперболического типа. Иногда пользуются другой канонической формой уравнения гиперболического типа. Сделаем в уравнении (3.13) замену переменных по закону $\xi = t + \tau$, $\eta = t - \tau$, где t и τ — новые переменные. В результате этого преобразования уравнение (3.13) примет вид

$$\frac{\partial^2 u}{\partial t^2} - \frac{\partial^2 u}{\partial \tau^2} = \tilde{\Phi},$$

где $\tilde{\Phi} = 4\bar{\Phi}$. (Проверьте самостоятельно!)

Уравнение параболического типа

В этом случае уравнения (3.11) и (3.12) совпадают, и мы получаем один общий интеграл $\varphi(x, y) = \text{const}$, определяющий одно семейство характеристик. Тогда можно положить $\xi = \varphi(x, y)$, $\eta = \psi(x, y)$, где $\psi(x, y)$ — любая функция, независимая от функции $\varphi(x, y)$, лишь бы она была дифференцируема нужное число раз. Очевидно, при выбранной замене переменных коэффициент \bar{A} в уравнении (3.8) обращается в нуль, то есть

$$\bar{A} = A \left(\frac{\partial \xi}{\partial x} \right)^2 + 2B \frac{\partial \xi}{\partial x} \frac{\partial \xi}{\partial y} + C \left(\frac{\partial \xi}{\partial y} \right)^2 = 0.$$

С учетом того, что $AC - B^2 = 0$ или $B = \sqrt{A}\sqrt{C}$, последнее равенство можно переписать в виде

$$A \left(\frac{\partial \xi}{\partial x} \right)^2 + 2B \frac{\partial \xi}{\partial x} \frac{\partial \xi}{\partial y} + C \left(\frac{\partial \xi}{\partial y} \right)^2 = \left(\sqrt{A} \frac{\partial \xi}{\partial x} + \sqrt{C} \frac{\partial \xi}{\partial y} \right)^2 = 0.$$

Это поможет доказать, что коэффициент $\bar{B} = 0$. Действительно,

$$\begin{aligned} \bar{B} &= A \frac{\partial \xi}{\partial x} \frac{\partial \eta}{\partial x} + B \left(\frac{\partial \xi}{\partial x} \frac{\partial \eta}{\partial y} + \frac{\partial \eta}{\partial x} \frac{\partial \xi}{\partial y} \right) + C \frac{\partial \xi}{\partial y} \frac{\partial \eta}{\partial y} = \\ &= A \frac{\partial \xi}{\partial x} \frac{\partial \eta}{\partial x} + \sqrt{A}\sqrt{C} \frac{\partial \xi}{\partial x} \frac{\partial \eta}{\partial y} + \sqrt{A}\sqrt{C} \frac{\partial \eta}{\partial x} \frac{\partial \xi}{\partial y} + C \frac{\partial \xi}{\partial y} \frac{\partial \eta}{\partial y} = \\ &= \sqrt{A} \frac{\partial \xi}{\partial x} \left(\sqrt{A} \frac{\partial \eta}{\partial x} + \sqrt{C} \frac{\partial \eta}{\partial y} \right) + \sqrt{C} \frac{\partial \xi}{\partial y} \left(\sqrt{A} \frac{\partial \eta}{\partial x} + \sqrt{C} \frac{\partial \eta}{\partial y} \right) = \end{aligned}$$

$$= \left(\sqrt{A} \frac{\partial \xi}{\partial x} + \sqrt{C} \frac{\partial \xi}{\partial y} \right) \left(\sqrt{A} \frac{\partial \eta}{\partial x} + \sqrt{C} \frac{\partial \eta}{\partial y} \right) = 0.$$

Итак, уравнение (3.8) принимает вид

$$\bar{C} \frac{\partial^2 u}{\partial \eta^2} + \Phi \left(\xi, \eta, u, \frac{\partial u}{\partial \xi}, \frac{\partial u}{\partial \eta} \right) = 0.$$

После деления на \bar{C} окончательно получим

$$\frac{\partial^2 u}{\partial \eta^2} = \tilde{\Phi} \left(\xi, \eta, u, \frac{\partial u}{\partial \xi}, \frac{\partial u}{\partial \eta} \right). \quad (3.14)$$

Уравнение (3.14) называется *канонической формой уравнения параболического типа*. Интересно отметить, что если правая часть уравнения (3.14) не содержит производной $\partial u / \partial \xi$, то оно становится обыкновенным дифференциальным уравнением, где роль параметра играет ξ .

Уравнение эллиптического типа

Правые части уравнений (3.11) и (3.12) комплексно сопряжены. Пусть $\varphi(x, y)$ — комплексный интеграл уравнения (3.11). Тогда $\varphi^*(x, y) = \text{const}$ интеграл уравнения (3.12), где $\varphi^*(x, y)$ — функция, комплексно сопряженная с функцией $\varphi(x, y)$. Если теперь перейти к комплексным переменным $\xi = \varphi(x, y)$, $\eta = \varphi^*(x, y)$, то согласно общей теории уравнение (3.6) приведет к виду

$$\frac{\partial^2 u}{\partial \xi \partial \eta} = \Phi \left(\xi, \eta, u, \frac{\partial u}{\partial \xi}, \frac{\partial u}{\partial \eta} \right),$$

то есть точно к такому же виду, как и гиперболическое уравнение. Чтобы остаться в действительной области, сделаем еще одну замену переменных

$$\alpha = \frac{\varphi + \varphi^*}{2}, \quad \beta = \frac{\varphi - \varphi^*}{2i}, \quad i = \sqrt{-1},$$

где α и β — новые переменные. Тогда $\xi = \alpha + i\beta$, $\eta = \alpha - i\beta$. Нетрудно показать, что при такой замене переменных $\bar{A} = \bar{C}$, $\bar{B} = 0$. Таким образом, уравнение (3.6) приводится к виду

$$\frac{\partial^2 u}{\partial \alpha^2} + \frac{\partial^2 u}{\partial \beta^2} = \Theta \left(\alpha, \beta, u, \frac{\partial u}{\partial \alpha}, \frac{\partial u}{\partial \beta} \right),$$

который называется *каноническим видом уравнения эллиптического типа*.

Понятие об общем интеграле уравнения в частных производных

Как мы знаем, для обыкновенного дифференциального уравнения n -го порядка вся совокупность решений (за исключением «особых» решений) представляется функцией от независимой переменной x , а также от n произвольных постоянных интегрирования C_1, C_2, \dots, C_n . Наоборот, для любого семейства функций $u = \varphi(x, C_1, C_2, \dots, C_n)$, зависящего от n параметров, существует дифференциальное уравнение n -го порядка, решение которого $u = \varphi$ получается исключением параметров C_1, C_2, \dots, C_n из уравнения $u = \varphi(x, C_1, C_2, \dots, C_n)$ и $n - 1$ уравнений

$$\begin{aligned} u' &= \varphi'(x, C_1, C_2, \dots, C_n), \\ &\dots\dots\dots \\ u^{(n-1)} &= \varphi^{(n-1)}(x, C_1, C_2, \dots, C_n). \end{aligned}$$

Для дифференциальных уравнений с частными производными дело обстоит сложнее. Здесь тоже можно искать всю совокупность решений, или «общее решение», то есть такое решение, которое дает любое частное решение после того, как фиксированы некоторые «произвольные» элементы (опять за исключением некоторых «особых» решений). В случае дифференциальных уравнений с частными производными эти произвольные элементы уже не могут быть постоянными интегрирования, а должны содержать произвольные функции; вообще говоря, число этих произвольных функций равно порядку дифференциального уравнения. Число аргументов этих произвольных функций на единицу меньше числа аргументов решения u .

Пожалуй, самое яркое отличие здесь заключается в том, что задачи, приводящие к дифференциальным уравнениям в частных производных, почти нигде не встречаются в такой форме, чтобы искомым являлось общее решение. Мало того, если вы располагаете таким решением, то знание его, за редким исключением, почти ничего не дает для решения основной частной задачи. Ведь общее решение уравнения в частных производных зависит от произвольных функций и для нахождения частного решения придется рассматривать систему сложных функциональных соотношений, так что отыскание произвольных функций часто оказывается практически невозможным.

Примеры

Рассмотрим определение общих интегралов некоторых дифференциальных уравнений в частных производных. Следует отметить, что этот вопрос тесно связан с преобразованием дифференциального уравнения в частных производных к каноническому виду.

Пример 1. Пусть $u = u(x, y)$. Дифференциальное уравнение

$$\frac{\partial u}{\partial y} = 0$$

для функции $u(x, y)$ означает, что u не зависит от y ; следовательно, $u = w(x)$, где $w(x)$ — произвольная функция x .

Пример 2. Для уравнения

$$\frac{\partial^2 u}{\partial x \partial y} = 0,$$

где $u = u(x, y)$, можно сразу получить общее решение вида $u = w(x) + v(y)$. Здесь $w(x)$ и $v(y)$ произвольные дифференцируемые функции. Действительно,

$$\frac{\partial^2 u}{\partial x \partial y} = 0 \Rightarrow \frac{\partial u}{\partial y} = f(y),$$

т.е. частная производная по y не зависит от x . Интегрируя последнее равенство, находим

$$u = \int f(y) dy + w(x).$$

Теперь остается только обозначить первообразную через $v(y)$.

Пример 3. Пусть $u = u(x, y)$. Рассмотрим дифференциальное уравнение

$$\frac{\partial u}{\partial x} = \frac{\partial u}{\partial y}.$$

Найдем общий интеграл этого уравнения. Для этого сделаем замену переменных

$$x + y = \xi, \quad x - y = \eta.$$

Тогда

$$\frac{\partial u}{\partial x} = \frac{\partial u}{\partial \xi} \frac{\partial \xi}{\partial x} + \frac{\partial u}{\partial \eta} \frac{\partial \eta}{\partial x} = \frac{\partial u}{\partial \xi} + \frac{\partial u}{\partial \eta},$$

$$\frac{\partial u}{\partial y} = \frac{\partial u}{\partial \xi} \frac{\partial \xi}{\partial y} + \frac{\partial u}{\partial \eta} \frac{\partial \eta}{\partial y} = \frac{\partial u}{\partial \xi} - \frac{\partial u}{\partial \eta}$$

и уравнение принимает вид

$$\frac{\partial u}{\partial \eta} = 0.$$

Значит $u = f(\xi)$. Возвращаясь к старым переменным, получим $u = f(x + y)$, причем f — любая дифференцируемая функция.

Пример 4. Пусть $u = u(x, y)$. Рассмотрим дифференциальное уравнение

$$x \frac{\partial u}{\partial x} + y \frac{\partial u}{\partial y} = 0.$$

Сделаем замену переменных

$$\frac{y}{x} = \xi, \quad y = \eta.$$

Тогда

$$\frac{\partial u}{\partial x} = -\frac{\partial u}{\partial \xi} \frac{y}{x^2} + \frac{\partial u}{\partial \eta} \cdot 0 = -\frac{\partial u}{\partial \xi} \frac{y}{x^2},$$

$$\frac{\partial u}{\partial y} = \frac{\partial u}{\partial \xi} \frac{1}{x} + \frac{\partial u}{\partial \eta} \cdot 1$$

и уравнение принимает вид

$$\frac{\partial u}{\partial \eta} = 0.$$

Значит $u = f(\xi)$. Следовательно, общий интеграл уравнения будет $u = f(y/x)$, причем f — любая дифференцируемая функция.

Пример 5. Пусть $u = u(x, y)$. Рассмотрим дифференциальное уравнение

$$\frac{\partial^2 u}{\partial x^2} - \frac{\partial^2 u}{\partial y^2} = 0.$$

Сделаем замену переменных

$$x + y = \xi, \quad x - y = \eta.$$

Тогда

$$\frac{\partial u}{\partial x} = \frac{\partial u}{\partial \xi} + \frac{\partial u}{\partial \eta},$$

$$\frac{\partial u}{\partial y} = \frac{\partial u}{\partial \xi} - \frac{\partial u}{\partial \eta},$$

$$\frac{\partial^2 u}{\partial x^2} = \frac{\partial}{\partial x} \left(\frac{\partial u}{\partial x} \right) = \frac{\partial}{\partial \xi} \left(\frac{\partial u}{\partial \xi} + \frac{\partial u}{\partial \eta} \right) + \frac{\partial}{\partial \eta} \left(\frac{\partial u}{\partial \xi} + \frac{\partial u}{\partial \eta} \right) = \frac{\partial^2 u}{\partial \xi^2} + 2 \frac{\partial^2 u}{\partial \xi \partial \eta} + \frac{\partial^2 u}{\partial \eta^2},$$

$$\frac{\partial^2 u}{\partial y^2} = \frac{\partial}{\partial y} \left(\frac{\partial u}{\partial y} \right) = \frac{\partial^2 u}{\partial \xi^2} - 2 \frac{\partial^2 u}{\partial \xi \partial \eta} + \frac{\partial^2 u}{\partial \eta^2},$$

и уравнение принимает вид

$$4 \frac{\partial^2 u}{\partial \xi \partial \eta} = 0.$$

Значит $u = F(\xi) + G(\eta)$. Следовательно, общий интеграл уравнения будет $u = F(x + y) + G(x - y)$, где F и G — произвольные дважды дифференцируемые функции.

Пример 6. Пусть $u = u(x, y)$. Общий интеграл дифференциального уравнения

$$\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} = 0$$

дается функцией $u = F(x + iy) + G(x - iy)$, где F и G — произвольные дважды дифференцируемые функции, $i = \sqrt{-1}$.

Применение общего интеграла к решению некоторых задач математической физики

1. Пусть некоторый процесс описывается одномерным волновым уравнением

$$\frac{\partial^2 u}{\partial x^2} - \frac{1}{v^2} \frac{\partial^2 u}{\partial t^2} = 0.$$

Обозначим $vt = y$, тогда уравнение примет вид

$$\frac{\partial^2 u}{\partial x^2} - \frac{\partial^2 u}{\partial y^2} = 0.$$

Следовательно, общий интеграл волнового уравнения будет

$$u = F(x + vt) + G(x - vt).$$

Причем здесь $F(x + vt)$ — плоская волна, распространяющаяся влево от начала координат со скоростью v , форма которой (профиль) определяется функцией F . Такая волна называется обратной волной. Функция $G(x - vt)$ описывает плоскую волну, распространяющуюся вправо от начала координат со скоростью v — прямая волна. Таким образом, рассматриваемый процесс представляет собой результат наложения двух волн.

2. Рассмотрим процесс, описываемый уравнением

$$\frac{1}{r^2} \frac{\partial}{\partial r} \left(r^2 \frac{\partial u}{\partial r} \right) - \frac{1}{v^2} \frac{\partial^2 u}{\partial t^2} = 0.$$

Это трехмерное волновое уравнение, записанное в сферических координатах в предположении, что процесс не зависит от угловых координат (радиальные колебания).

Сделаем замену переменных $u = \frac{w}{r}$, тогда $\frac{\partial u}{\partial r} = \frac{1}{r} \frac{\partial w}{\partial r} - \frac{1}{r^2} w$, $r^2 \frac{\partial u}{\partial r} = r \frac{\partial w}{\partial r} - w$,
 $\frac{1}{r^2} \frac{\partial}{\partial r} \left(r^2 \frac{\partial u}{\partial r} \right) = \frac{1}{r} \frac{\partial^2 w}{\partial r^2}$, $\frac{\partial^2 u}{\partial t^2} = \frac{1}{r} \frac{\partial^2 w}{\partial t^2}$. В результате получим уравнение

$$\frac{\partial^2 w}{\partial r^2} - \frac{1}{v^2} \frac{\partial^2 w}{\partial t^2} = 0.$$

Откуда

$$w = F(r + vt) + G(r - vt) \Rightarrow u = \frac{F(r + vt)}{r} + \frac{G(r - vt)}{r}.$$

Первое слагаемое представляет собой сферическую волну, распространяющуюся из бесконечности в начало координат со скоростью v (обратная волна), второе слагаемое — сферическая волна, распространяющаяся со скоростью v от начала координат в бесконечность (прямая волна). Таким образом, и в этом случае общее решение уравнения представляет собой наложение прямой и обратной волны.

Колебания неограниченной струны. Метод Даламбера

Пусть требуется найти функцию $u = u(x, t)$, непрерывную в замкнутой области $\{-\infty < x < \infty, t \geq 0\}$, удовлетворяющую уравнению

$$\frac{\partial^2 u}{\partial x^2} - \frac{1}{v^2} \frac{\partial^2 u}{\partial t^2} = 0, \quad -\infty < x < \infty, t > 0, \quad (3.15)$$

и начальным условиям

$$u(x, 0) = \Phi(x), \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = \Psi(x), \quad -\infty < x < \infty. \quad (3.16)$$

Общий интеграл уравнения имеет вид

$$u = F(x + vt) + G(x - vt) \Rightarrow \frac{\partial u}{\partial t} = vF'(x + vt) - vG'(x - vt).$$

При $t = 0$ будем иметь

$$F(x) + G(x) = \Phi(x), \quad F'(x) - G'(x) = \frac{1}{v} \Psi(x).$$

Откуда находим

$$2F'(x) = \Phi'(x) + \frac{1}{v} \Psi(x), \quad 2G'(x) = \Phi'(x) - \frac{1}{v} \Psi(x),$$

или

$$F(x) = \frac{1}{2}\Phi(x) + \frac{1}{2v} \int_0^x \Psi(\xi) d\xi + c_1, \quad G(x) = \frac{1}{2}\Phi(x) - \frac{1}{2v} \int_0^x \Psi(\xi) d\xi + c_2.$$

После подстановки в выражение для u получим

$$u = \frac{1}{2}\Phi(x+vt) + \frac{1}{2v} \int_0^{x+vt} \Psi(\xi) d\xi + \frac{1}{2}\Phi(x-vt) - \frac{1}{2v} \int_0^{x-vt} \Psi(\xi) d\xi + c,$$

где $c = c_1 + c_2$. Легко видеть, что $c = 0$. Это следует из начального условия $u(x, 0) = \Phi(x)$.

Итак,

$$u = \frac{1}{2}[\Phi(x+vt) + \Phi(x-vt)] + \frac{1}{2v} \int_{x-vt}^{x+vt} \Psi(\xi) d\xi. \quad (3.17)$$

Формулу (3.17) называют *формулой Даламбера* решения задачи Коши для волнового уравнения (3.15).

До сих пор мы не накладывали никаких ограничений на функции $\Phi(x)$ и $\Psi(x)$, фигурирующие в начальных условиях (3.16). Формула (3.17) будет давать решение задачи Коши в том случае, если функция $\Phi(x)$ обладает производными первого и второго порядков, а функция $\Psi(x)$ — производной первого порядка. К этому выводу мы придем, когда захотим убедиться в правильности полученного решения путем подстановки его в уравнение (3.15). (Проверьте!)

Решение u зависит от начальной функции $\Psi(x)$ только в интервале $(x-vt, x+vt)$. Значения функции $\Psi(x)$ за пределами этого интервала не играют роли при определении решения u .

Метод Даламбера допускает распространение и на более сложные случаи.

Во многих задачах математической физики, с которыми мы встретимся, существование решения устанавливается лишь при значительных ограничениях на краевые условия. Мы введем одно понятие, которое позволит нам не заниматься далее этими вопросами.

Рассмотрим снова задачу Коши для волнового уравнения (3.15) при начальных условиях (3.16). Как мы только что показали, решением этой задачи будет функция, определяемая по формуле (3.17). Эта формула дает классическое решение уравнения (3.15) только в предположении, что $\Phi(x)$ имеет непрерывные производные до второго порядка включительно, а $\Psi(x)$ — до первого.

Задача Коши (3.15)–(3.16) поставлена *корректно*. Действительно, полученное решение единственно, что следует из способа вывода формулы (3.17). Несомненно далее непрерывная зависимость решения (3.17) от начальных данных. В самом деле,

для любого $\varepsilon > 0$ можно указать такое $\delta > 0$, что если заменить $\Phi(x)$ и $\Psi(x)$ на $\tilde{\Phi}(x)$ и $\tilde{\Psi}(x)$ так, что

$$|\Phi(x) - \tilde{\Phi}(x)| < \delta, \quad |\Psi(x) - \tilde{\Psi}(x)| < \delta, \quad -\infty < x < \infty,$$

то разность между новым решением $\tilde{u}(x, t)$ и первоначальным $u(x, t)$ будет по абсолютной величине меньше ε на любом конечном отрезке времени. Это утверждение легко следует из формулы (3.17).

При решении конкретных физических задач может оказаться, что функции $\Phi(x)$ и $\Psi(x)$ не удовлетворяют указанным условиям. Тогда нельзя утверждать, что существует решение задачи Коши. В этом случае вводят так называемые *обобщенные решения* задачи Коши.

Будем называть *обобщенным решением задачи Коши* для уравнения (3.15) при начальных условиях (3.16) функцию $u(x, t)$, являющуюся пределом равномерно сходящейся последовательности решений $u_n(x, t)$ уравнения (3.15) при начальных условиях

$$u_n(x, 0) = \Phi_n(x), \quad \left. \frac{\partial u_n}{\partial t} \right|_{t=0} = \Psi_n(x), \quad -\infty < x < \infty,$$

если последовательность функций $\Phi_n(x)$, имеющих непрерывные вторые производные, сходится равномерно к $\Phi(x)$, а последовательность функций $\Psi_n(x)$, имеющих непрерывные первые производные, сходится равномерно к $\Psi(x)$.

Нетрудно доказать существование и единственность обобщенного решения задачи Коши для уравнения (3.15) при любых непрерывных функциях $\Phi(x)$ и $\Psi(x)$. Это обобщенное решение также дается формулой (3.17).

Введение обобщенных решений уравнения (3.15) естественно тем, что, во-первых, для существования обычного (классического) решения задачи Коши приходится на заданные функции $\Phi(x)$ и $\Psi(x)$ налагать весьма жесткие условия гладкости, в то время как для существования обобщенных решений такой гладкости от заданных функций не требуется. Во-вторых, функции $\Phi(x)$ и $\Psi(x)$ в конкретных задачах физики известны нам только приближенно. Поэтому соответствующая функция $u(x, t)$, даваемая формулой (3.17), также является только некоторым приближением к точному решению поставленной задачи.

Следовательно, совершенно безразлично, является ли это приближение обычным или обобщенным решением задачи Коши. Важно, что оно будет мало отличаться от истинного решения, если только функции $\Phi(x)$ и $\Psi(x)$ равномерно мало отличаются от истинных начальных значений $u(x, 0)$ и $\frac{\partial u(x, 0)}{\partial t}$.

Мы говорили сейчас об обобщенном решении задачи Коши для одномерного волнового уравнения. То же самое относится к общему волновому уравнению, к уравнению теплопроводности и к уравнению Пуассона.

Обобщение формулы Даламбера на многомерные задачи

Классической задачей Коши для волнового уравнения называется задача [2] о нахождении функции $u(x, t)$ класса $C^{(2)}(t > 0) \cap C^{(1)}(t \geq 0)$, удовлетворяющей при $t > 0$ волновому уравнению

$$\frac{\partial^2 u}{\partial t^2} = v^2 \Delta u + f(x, t) \quad (3.18)$$

и начальным условиям

$$u|_{t=0} = \Phi(x), \quad \frac{\partial u}{\partial t}|_{t=0} = \Psi(x), \quad (3.19)$$

где f, Φ, Ψ — заданные функции и $x = (x_1, x_2, \dots, x_n) \in R^n$.

Если выполняются условия

$$\begin{aligned} f &\in C^{(1)}(t \geq 0), \quad \Phi \in C^{(2)}(R^1), \quad \Psi \in C^{(1)}(R^1), \quad n = 1, \\ f &\in C^{(2)}(t \geq 0), \quad \Phi \in C^{(3)}(R^n), \quad \Psi \in C^{(2)}(R^2), \quad n = 2, 3, \end{aligned} \quad (3.20)$$

то решение задачи Коши (3.18)–(3.19) существует, единственно и выражается формулой Даламбера при $n = 1$

$$u = \frac{1}{2} [\Phi(x + vt) + \Phi(x - vt)] + \frac{1}{2v} \int_{x-vt}^{x+vt} \Psi(\xi) d\xi + \frac{1}{2v} \int_0^t \int_{x-v(t-\tau)}^{x+v(t-\tau)} f(\xi, \tau) d\xi d\tau, \quad (3.21)$$

формулой Пуассона при $n = 2$

$$\begin{aligned} u(x, t) &= \frac{1}{2\pi v} \int_0^t \int_{|\xi-x| < v(t-\tau)} \frac{f(\xi, \tau) d\xi d\tau}{\sqrt{v^2(t-\tau)^2 - |\xi-x|^2}} + \\ &+ \frac{1}{2\pi v} \int_{|\xi-x| < vt} \frac{\Psi(\xi) d\xi}{\sqrt{v^2 t^2 - |\xi-x|^2}} + \frac{1}{2\pi v} \frac{\partial}{\partial t} \int_{|\xi-x| < vt} \frac{\Phi(\xi) d\xi}{\sqrt{v^2 t^2 - |\xi-x|^2}}, \end{aligned} \quad (3.22)$$

формулой Кирхгофа при $n = 3$

$$u(x, t) = \frac{1}{4\pi v^2} \int_{|\xi-x| < vt} \frac{f\left(\xi, t - \frac{|\xi-x|}{v}\right) d\xi}{|\xi-x|} +$$

$$+ \frac{1}{4\pi v^2 t} \int_{|\xi-x|=vt} \Psi(\xi) dS + \frac{1}{4\pi v^2} \frac{\partial}{\partial t} \left[\frac{1}{t} \int_{|\xi-x|=vt} \Phi(\xi) dS \right]. \quad (3.23)$$

Примеры решения типовых задач

Задача 3.1. Привести к каноническому виду уравнение

$$x^2 \frac{\partial^2 u}{\partial x^2} - y^2 \frac{\partial^2 u}{\partial y^2} = 0, \quad x > 0, \quad y > 0$$

и найти его общее решение.

Решение. Определим сначала тип уравнения. Имеем

$$A = x^2, \quad B = 0, \quad C = -y^2 \Rightarrow \Delta = AC - B^2 = -x^2 y^2 < 0.$$

Следовательно, тип уравнения — гиперболический.

Составим характеристическое уравнение

$$x^2 dy^2 - y^2 dx^2 = 0.$$

Это уравнение распадается на два уравнения с разделяющимися переменными

$$xdy - ydx = 0, \quad xdy + ydx = 0.$$

Откуда находим

$$xy = C_1, \quad \frac{y}{x} = C_2.$$

Таким образом, прямые $y = C_2 x$ и гиперболы $y = \frac{C_1}{x}$ являются характеристиками заданного уравнения.

Преобразуем теперь заданное уравнение к каноническому виду. Для этого, согласно общей теории, введем новые переменные по формулам

$$\xi = xy, \quad \eta = \frac{y}{x}.$$

Найдем $\frac{\partial^2 u}{\partial x^2}$ и $\frac{\partial^2 u}{\partial y^2}$:

$$\frac{\partial^2 u}{\partial x^2} = y^2 \frac{\partial^2 u}{\partial \xi^2} - 2 \frac{y^2}{x^2} \frac{\partial^2 u}{\partial \xi \partial \eta} + \frac{y^2}{x^4} \frac{\partial^2 u}{\partial \eta^2} + 2 \frac{y}{x^3} \frac{\partial u}{\partial \eta};$$

$$\frac{\partial^2 u}{\partial y^2} = x^2 \frac{\partial^2 u}{\partial \xi^2} + 2 \frac{\partial^2 u}{\partial \xi \partial \eta} + \frac{1}{x^2} \frac{\partial^2 u}{\partial \eta^2}.$$

Подставляя эти значения вторых производных в исходное уравнение, окончательно получим

$$\frac{\partial^2 u}{\partial \xi \partial \eta} - \frac{1}{2\xi} \frac{\partial u}{\partial \eta} = 0, \quad \xi > 0, \quad \eta > 0.$$

Последнее уравнение можно проинтегрировать. Для этого перепишем его в виде

$$\frac{\partial}{\partial \xi} \left(\frac{\partial u}{\partial \eta} \right) - \frac{1}{2\xi} \frac{\partial u}{\partial \eta} = 0.$$

Введем теперь новую неизвестную функцию $v(\xi, \eta) = \frac{\partial u}{\partial \eta}$. Будем иметь

$$\frac{\partial v}{\partial \xi} - \frac{v}{2\xi} = 0 \Rightarrow \frac{dv}{v} = \frac{d\xi}{2\xi} \Rightarrow \ln|v| = \frac{1}{2} \ln|\xi| + \ln|f(\eta)|,$$

где $f(\eta)$ — произвольная функция η . После потенцирования получим

$$v = f(\eta) \sqrt{\xi}.$$

Далее,

$$\frac{\partial u}{\partial \eta} = v(\xi, \eta) \Rightarrow \frac{\partial u}{\partial \eta} = f(\eta) \sqrt{\xi} \Rightarrow u = \sqrt{\xi} \int f(\eta) d\eta + \psi(\xi),$$

где $\psi(\xi)$ — произвольная функция ξ . Обозначив

$$\varphi(\eta) = \int f(\eta) d\eta,$$

окончательно получим

$$u = \sqrt{\xi} \varphi(\eta) + \psi(\xi),$$

где $\psi(\xi)$ и $\varphi(\eta)$ — произвольные функции ξ и η соответственно.

Возвращаясь к старым переменным x и y , получим

$$u = \sqrt{xy} \varphi\left(\frac{y}{x}\right) + \psi(xy).$$

Задача 3.2. Привести к каноническому виду уравнение

$$x^2 \frac{\partial^2 u}{\partial x^2} + 2xy \frac{\partial^2 u}{\partial x \partial y} + y^2 \frac{\partial^2 u}{\partial y^2} = 0, \quad x > 0$$

и найти его общее решение.

Решение. Определим тип уравнения. Имеем

$$A = x^2, B = xy, C = y^2 \Rightarrow \Delta = AC - B^2 = x^2y^2 - x^2y^2 = 0.$$

Следовательно, тип уравнения — параболический.

Составим характеристическое уравнение

$$x^2 dy^2 - 2xy dx dy + y^2 dx^2 = 0 \Rightarrow (xdy - ydx)^2 = 0 \Rightarrow xdy - ydx = 0.$$

Разделяя переменные и интегрируя, получим одно семейство характеристик

$$\frac{y}{x} = C.$$

Это — семейство прямых. Новые переменные вводим по формулам

$$\xi = \frac{y}{x}, \eta = y.$$

Выбор второй функции сделан с учетом того, чтобы она была наиболее простой, но так, чтобы функциональный определитель Остроградского для преобразования переменных был отличен от нуля.

Находим теперь

$$\frac{\partial^2 u}{\partial x^2} = \frac{y^2}{x^4} \frac{\partial^2 u}{\partial \xi^2} + 2 \frac{y}{x^3} \frac{\partial u}{\partial \xi};$$

$$\frac{\partial^2 u}{\partial x \partial y} = -\frac{y}{x^3} \frac{\partial^2 u}{\partial \xi^2} - \frac{y}{x^2} \frac{\partial^2 u}{\partial \xi \partial \eta} - \frac{1}{x^2} \frac{\partial u}{\partial \xi};$$

$$\frac{\partial^2 u}{\partial y^2} = \frac{1}{x^2} \frac{\partial^2 u}{\partial \xi^2} + \frac{2}{x} \frac{\partial^2 u}{\partial \xi \partial \eta} + \frac{\partial^2 u}{\partial \eta^2}.$$

Подставляя полученные выражения производных в заданное уравнение, окончательно получим каноническую форму этого уравнения

$$\frac{\partial^2 u}{\partial \eta^2} = 0.$$

Полученное уравнение легко интегрируется. Будем иметь

$$\frac{\partial u}{\partial \eta} = f(\xi) \Rightarrow u = \int f(\xi) d\eta + \varphi(\xi)$$

или

$$u = f(\xi)\eta + \varphi(\xi),$$

где f и φ — произвольные функции.

Возвращаясь к старым переменным, получим общее решение заданного уравнения

$$u = f\left(\frac{y}{x}\right)y + \varphi\left(\frac{y}{x}\right).$$

Задача 3.3. Привести к каноническому виду уравнение

$$\frac{\partial^2 u}{\partial x^2} - 4 \frac{\partial^2 u}{\partial x \partial y} + 5 \frac{\partial^2 u}{\partial y^2} = 0.$$

Решение. Определим тип уравнения. Имеем

$$A = 1, B = -2, C = 5 \Rightarrow \Delta = AC - B^2 = 5 - 4 = 1 > 0.$$

Следовательно, тип уравнения — эллиптический.

Составим характеристическое уравнение

$$dy^2 + 4dx dy + 5dx^2 = 0.$$

Это уравнение распадается на два следующих уравнения

$$y' = -2 + i, \quad y' = -2 - i.$$

Интегрируя их, найдем два общих интеграла характеристического уравнения

$$y = (-2 + i)x + C_1, \quad y = (-2 - i)x + C_2.$$

Введем обозначения

$$\varphi = y - (-2 + i)x, \quad \varphi^* = y + (2 + i)x,$$

$$\alpha = \frac{\varphi + \varphi^*}{2} = y + 2x, \quad \beta = \frac{\varphi - \varphi^*}{2i} = -x.$$

Согласно общей теории нужно ввести новые переменные α и β по формулам

$$\xi = \alpha + i\beta = y + 2x - ix, \quad \eta = \alpha - i\beta = y + 2x + ix.$$

Вычисляя производные как в уравнении (3.8), после всех преобразований окончательно получим канонический вид заданного уравнения

$$\frac{\partial^2 u}{\partial \xi^2} + \frac{\partial^2 u}{\partial \eta^2} = 0.$$

(Проверьте!)

Задача 3.4. Привести к каноническому виду уравнение

$$\frac{\partial^2 u}{\partial x^2} - 6 \frac{\partial^2 u}{\partial x \partial y} + 10 \frac{\partial^2 u}{\partial y^2} + \frac{\partial u}{\partial x} - 3 \frac{\partial u}{\partial y} = 0.$$

Решение. Воспользуемся для решения задачи системой аналитических вычислений Maple. Будем рассматривать уравнение общего вида

$$a_1 \frac{\partial^2 u}{\partial x^2} + a_2 \frac{\partial^2 u}{\partial x \partial y} + a_3 \frac{\partial^2 u}{\partial y^2} + a_4 \frac{\partial u}{\partial x} + a_5 \frac{\partial u}{\partial y} = 0.$$

Зададим коэффициенты нашего уравнения

```
> a := 1, -6, 10, 1, -3;
```

$$a := 1, -6, 10, 1, -3$$

и само уравнение

```
> equ:=a[1]*diff(u(x,y),x,x)+a[2]*diff(u(x,y),x,y)+
a[3]*diff(u(x,y),y,y)+a[4]*diff(u(x,y),x)+
a[5]*diff(u(x,y),y)=0;
```

$$equ := \left(\frac{\partial^2}{\partial x^2} u(x, y) \right) - 6 \left(\frac{\partial^2}{\partial y \partial x} u(x, y) \right) + 10 \left(\frac{\partial^2}{\partial y^2} u(x, y) \right) + \left(\frac{\partial}{\partial x} u(x, y) \right) - 3 \left(\frac{\partial}{\partial y} u(x, y) \right) = 0$$

Вычисляем матрицу старших коэффициентов и ее определитель. Для этого воспользуемся командой `matrix` из пакета `linalg`. Формат этой команды — `matrix(m,n,iv)`. Здесь m, n — положительные целые числа, определяющие количество строк и столбцов матрицы; iv — список элементов матрицы. Для формирования списка элементов мы используем команду `coeff(p,x)`. Эта команда извлекает коэффициенты многочлена p относительно x . Итак, определяем многочлен относительно производных и формируем матрицу коэффициентов. Затем вычисляем определитель этой матрицы

```
> eq:=lhs(equ);
```

$$eq := \left(\frac{\partial^2}{\partial x^2} u(x, y) \right) - 6 \left(\frac{\partial^2}{\partial y \partial x} u(x, y) \right) + 10 \left(\frac{\partial^2}{\partial y^2} u(x, y) \right) + \left(\frac{\partial}{\partial x} u(x, y) \right) - 3 \left(\frac{\partial}{\partial y} u(x, y) \right)$$

```
> A:= linalg[matrix](2,2,[coeff(eq,diff(u(x,y),x,x)),
```

```
coeff(eq,diff(u(x,y),x,y))/2,coeff(eq,diff(u(x,y),x,y))/2,
```

```
coeff(eq,diff(u(x,y),y,y))]);
```

```
Delta:=simplify(linalg[det](A));
```

$$A := \begin{bmatrix} 1 & -3 \\ -3 & 10 \end{bmatrix}$$

$$\Delta := 1$$

Так как определитель матрицы старших коэффициентов $\Delta > 0$, то тип уравнения — эллиптический.

Формируем характеристическое уравнение и решаем его

```
> A[1,1]*z^2-2*A[1,2]*z+A[2,2]=0;
```

```
res1:=solve(A[1,1]*z^2-2*A[1,2]*z+A[2,2],z);
```

$$z^2 + 6z + 10 = 0$$

$$res1 := -3 + I, -3 - I$$

Таким образом, характеристическое уравнение распадается на два. Для формирования и решения этих уравнений воспользуемся следующей конструкцией

```
> res2 := {seq(dsolve(diff(y(x), x) = res1[i], y(x)), i = 1..2)};
      res2 := {y(x) = -(3 - I)x + _C1, y(x) = -(3 + I)x + _C1}
```

Здесь мы применили команду `seq(f, i=m..n)` для создания последовательности (в данном случае дифференциальных уравнений). Параметрами команды являются: `f` — любое выражение; `i` — имя; `m, n` — числовые значения.

Таким образом, мы получили две комплексные характеристики.

Выполняем замену переменных

```
> res2 := subs(y(x) = y, res2);
      res2 := {y = -(3 - I)x + _C1, y = -(3 + I)x + _C1}
> {seq(solve(res2[i], _C1), i = 1..nops(res2))};
      {y + 3x + xI, y + 3x - xI}
> itr := {xi = coeff(%[1], I), eta = %[1] - coeff(%[1], I) * I};
      itr := {η = y + 3x, ξ = x}
```

Теперь приводим заданное уравнение к канонической форме

```
> tr := solve(itr, {x, y});
      PDEtools[dchange](tr, eq, itr, [eta, xi], simplify) = 0;
      tr := {y = η - 3ξ, x = ξ}
```

$$\left(\frac{\partial^2}{\partial \eta^2} u(\eta, \xi) \right) + \left(\frac{\partial^2}{\partial \xi^2} u(\eta, \xi) \right) + \left(\frac{\partial}{\partial \xi} u(\eta, \xi) \right) = 0$$

Здесь мы воспользовались командой `dchange` из пакета `PDEtools`. Эта команда выполняет замену переменных в математических выражениях или процедурах. Формат команды: `dchange(tr, expr, itr, opts)`. Здесь `tr` — система (множество) уравнений, соответствующая преобразованию от старых переменных (в левых частях уравнений) к новым переменным (в правых частях уравнений); `expr` — алгебраическое выражение или процедура (это могут быть дифференциальные уравнения, в том числе и в частных производных, интегро-дифференциальные уравнения, интегралы, пределы и т.д.); `itr` — множество обратных уравнений преобразования; `opts` — дополнительные опции.

Итак, окончательно каноническая форма уравнения будет

$$\frac{\partial^2 u}{\partial \xi^2} + \frac{\partial^2 u}{\partial \eta^2} + \frac{\partial u}{\partial \xi} = 0.$$

Задача 3.5. Привести к каноническому виду уравнение

$$4 \frac{\partial^2 u}{\partial x^2} + 4 \frac{\partial^2 u}{\partial x \partial y} + \frac{\partial^2 u}{\partial y^2} - 2 \frac{\partial u}{\partial y} = 0.$$

Решение. Решаем задачу в системе аналитических вычислений Maple. Задаем уравнение

```
> a := 4, 4, 1, 0, -2;
```

```
a := 4, 4, 1, 0, -2
```

```
> equ:=a[1]*diff(u(x,y),x,x)+a[2]*diff(u(x,y),x,y)+
a[3]*diff(u(x,y),y,y)+a[4]*diff(u(x,y),x)+
a[5]*diff(u(x,y),y)=0;
```

$$equ := 4 \left(\frac{\partial^2}{\partial x^2} u(x, y) \right) + 4 \left(\frac{\partial^2}{\partial y \partial x} u(x, y) \right) + \left(\frac{\partial^2}{\partial y^2} u(x, y) \right) - 2 \left(\frac{\partial}{\partial y} u(x, y) \right) = 0$$

Вычисляем матрицу старших коэффициентов и ее определитель

```
> eq:=lhs(equ);
```

$$eq := 4 \left(\frac{\partial^2}{\partial x^2} u(x, y) \right) + 4 \left(\frac{\partial^2}{\partial y \partial x} u(x, y) \right) + \left(\frac{\partial^2}{\partial y^2} u(x, y) \right) - 2 \left(\frac{\partial}{\partial y} u(x, y) \right)$$

```
> A:= linalg[matrix](2,2,[coeff(eq,diff(u(x,y),x,x)),
coeff(eq,diff(u(x,y),x,y))/2,coeff(eq,diff(u(x,y),x,y))/2,
coeff(eq,diff(u(x,y),y,y))]);
```

```
Delta:=simplify(linalg[det](A));
```

$$A := \begin{bmatrix} 4 & 2 \\ 2 & 1 \end{bmatrix}$$

$$\Delta := 0$$

Так как определитель матрицы старших коэффициентов $\Delta = 0$, то тип уравнения — параболический.

Формируем характеристическое уравнение и решаем его

```
> A[1,1]*z^2-2*A[1,2]*z+A[2,2]=0;
```

```
res1:=solve(A[1,1]*z^2-2*A[1,2]*z+A[2,2],z);
```

$$4z^2 - 4z + 1 = 0$$

$$res1 := \frac{1}{2}, \frac{1}{2}$$

```
> subs(y=y(x),res1[1]);res2:=dsolve(diff(y(x),x)=%,y(x));
```

$$\frac{1}{2}$$

$$res2 := y(x) = \frac{x}{2} + _C1$$

Получили одно семейство характеристик. Вводим замену переменных

```
> res2:=subs (y (x)=y, res2) ;
```

$$res2 := y = \frac{x}{2} + _C1$$

```
> itr:={xi=solve (res2, \_C1) , eta=y} ;
```

$$itr := \{ \xi = y - \frac{x}{2}, \eta = y \}$$

Приводим заданное уравнение к каноническому виду

```
> tr:=solve (itr, {x, y}) ;
```

```
PDEtools[dchange] (tr, eq, itr, [eta, xi], simplify)=0;
```

$$tr := \{ y = \eta, x = -2 \xi + 2 \eta \}$$

$$\left(\frac{\partial^2}{\partial \eta^2} u(\eta, \xi) \right) - 2 \left(\frac{\partial}{\partial \xi} u(\eta, \xi) \right) - 2 \left(\frac{\partial}{\partial \eta} u(\eta, \xi) \right) = 0$$

Обратим внимание: если выбрать другую замену переменных

```
> itr:={xi=solve (res2, \_C1) , eta=x} ;
```

$$itr := \{ \xi = y - \frac{x}{2}, \eta = x \}$$

то придем к уравнению вида

```
> tr:=solve (itr, {x, y}) ;
```

```
PDEtools[dchange] (tr, eq, itr, [eta, xi], simplify)=0;
```

$$tr := \{ x = \eta, y = \xi + \frac{\eta}{2} \}$$

$$4 \left(\frac{\partial^2}{\partial \eta^2} u(\eta, \xi) \right) - 2 \left(\frac{\partial}{\partial \xi} u(\eta, \xi) \right) = 0$$

Последовательность действий по приведению уравнения к каноническому виду, которые мы выполняли, решая задачи 3.1 – 3.5, можно запрограммировать в виде специальной процедуры. Приведем такую программу, автоматизирующую преобразования по приведению линейных уравнений второго порядка с двумя независимыми переменными к канонической форме. Программа приводит к канонической форме уравнение следующего общего вида

$$a_1 \frac{\partial^2 u}{\partial x^2} + a_2 \frac{\partial^2 u}{\partial x \partial y} + a_3 \frac{\partial^2 u}{\partial y^2} + a_4 \frac{\partial u}{\partial x} + a_5 \frac{\partial u}{\partial y} = 0.$$

Формальным параметром программы служит именно это уравнение. Программа имеет два глобальных параметра — параметр `change_of_variables`, который

содержит формулы замены переменных и параметр `eq_can`, который содержит преобразованное уравнение. Эти переменные можно независимо использовать в последующей работе. Например, можно перейти к другой канонической форме гиперболического уравнения, содержащей смешанную вторую производную (программа преобразовывает квадратичную форму, соответствующую матрице старших коэффициентов, к сумме квадратов).

Еще одна особенность программы — классификация типов уравнений выполняется не по определителю квадратичной формы, он может зависеть от переменных, а по характеристикам, которые тоже однозначно определяют тип уравнения.

Алгоритм программы достаточно прозрачен. При необходимости ее можно легко модернизировать под те или иные нужды.

```
> canonical_form:=proc(equ)
  global change_of_variables,eq_can;
  local
  k,A,B,it,J,eq_n,r,tr,i,itr,eq_type,mu,eq,
  characteristics;
  eq:= lhs(equ):
  A:= linalg[matrix](2,2,[coeff(eq,diff(u(x,y),x,x)),
  coeff(eq,diff(u(x,y),x,y))/2,
  coeff(eq,diff(u(x,y),x,y))/2,
  coeff(eq,diff(u(x,y),y,y))]);
  simplify({solve(A[1,1]*z^2-2*A[1,2]*z+A[2,2],z)});
  simplify(%,power);simplify(%,power,symbolic);
  subs(y=y(x),%);
  if type(%[1],`*`) and has(%[1],y) and has(%[1],I) then
    mu:= 1/select(has,%[1],y);
    {seq(int(expand((diff(y(x),x)+%%[i])*mu),x)=_C1,
    i=1..nops(%%))};
  else
    {seq(dsolve(diff(y(x),x)=%[i],y(x)),i=1..nops(%))};
  end if;
  characteristics:=subs(y(x)=y,%);
  print(`Характеристики:`);print(characteristics);
  {seq(solve(characteristics[i],_C1),
  i=1..nops(characteristics))};
```

```

if nops(%)=1 and not hastype(%,nonreal) then
  itr:={xi=%[1],eta=x};      # либо eta=y
  eq_type:=`параболический тип`;
elif not hastype(%,nonreal) then
  itr:={xi=(%[1]+%[2])/2,eta=(%[2]-%[1])/2};
  eq_type:=`гиперболический тип`;
else
  itr:= {xi=coeff(%[1],I),eta=%[1]-coeff(%[1],I)*I};
  eq_type:=`эллиптический тип`;
end if;
itr:=simplify(itr);indets(itr);
select(has,%, [x,y]);
k:= solve(itr,%);
it:= convert(itr,list);
if has(%[1],eta) then it:= [%[2],%[1]] end if;
J:= jacobian([rhs(it[1]),rhs(it[2])],[x,y]);
B:= map(simplify,evalm(J*A&*transpose(J)));
eq_n:= B[1,1]*Diff(u,xi,xi)+2*B[1,2]*Diff(u,xi,eta)+
B[2,2]*Diff(u,eta,eta)+
simplify(subs(u(x,y)=rhs(it[1]),eq))*Diff(u,xi)+
simplify(subs(u(x,y)=rhs(it[2]),eq))*Diff(u,eta);
eq_n:= expand(eq_n);
k union map(x->1/lhs(x)=1/rhs(x),k);
subs(%,eq_n);
if eq_type=`параболический тип` then
  simplify(solve(%,Diff(u,eta,eta))):
  r:= simplify(subs(k,numer(%))/subs(k,denom(%)));
  if has(%, [x,y]) then
    tr:= solve(itr,{x,y});
    if has(%,RootOf) then tr:= allvalues(%) [1] end if;
    simplify(expand(subs(tr,r)));
  end if;
eq_can:= Diff(u,eta,eta)-%=0;

```

```

end if;
if eq_type=`гиперболический тип` then
  simplify(solve(eq_n,Diff(u,eta,eta))):
  r:= simplify(subs(k,numer(%))/subs(k,denom(%)));
  if has(%,[x,y]) then tr:= solve(itr,{x,y});
  if has(%,RootOf) then tr:= allvalues(%)[1] end if;
  simplify(expand(subs(tr,r)));
end if;
eq_can:= Diff(u,eta,eta)-Diff(u,xi,xi)-
simplify(%-Diff(u,xi,xi))=0;
end if;
if eq_type=`эллиптический тип` then
  simplify(solve(%,Diff(u,xi,xi))):
  r:= simplify(subs(k,numer(%))/subs(k,denom(%)));
  if has(%,[x,y]) then tr:= solve(itr,{x,y});
  if has(%,RootOf) then tr:= allvalues(%)[1] end if;
  map(simplify,subs(tr,r),trig,power,symbolic);
end if;
eq_can:= Diff(u,eta,eta)+Diff(u,xi,xi)-
simplify(%+Diff(u,eta,eta))=0;
end if;
change_of_variables:=itr;
print(eq_type);print(`Замена переменных:`);print(itr);
print(`Каноническая форма:`);
eq_can:= value(subs(u=u(eta,xi),eq_can));
RETURN(eq_can);
end proc:

```

При выборе замены переменных в случае уравнения параболического типа одна из переменных назначается произвольно: $\eta = x$, либо $\eta = y$ (лишь бы она была независима от ξ и дифференцируема нужное число раз). От этого выбора иногда зависит окончательный вид уравнения. Соответствующее место в программе закомментировано: `> itr:={xi=%[1],eta=x}; # либо eta=y`. Это означает, что в данном операторе можно использовать либо $\eta = x$, либо $\eta = y$.

Рассмотрим применение этой программы к решению задач.

Задача 3.6. Привести к каноническому виду уравнение

$$x^2 \frac{\partial^2 u}{\partial x^2} - 2x \frac{\partial^2 u}{\partial x \partial y} + \frac{\partial^2 u}{\partial y^2} = 0.$$

Решение. Для решения задачи воспользуемся программой `canonical_form`.

```
> a := x^2, -2*x, 1, 0, 0;
equ := a[1]*diff(u(x, y), x, x) + a[2]*diff(u(x, y), x, y) +
a[3]*diff(u(x, y), y, y) + a[4]*diff(u(x, y), x) +
a[5]*diff(u(x, y), y) = 0;
canonical_form(equ);
```

$$a := x^2, -2x, 1, 0, 0$$

$$equ := x^2 \left(\frac{\partial^2}{\partial x^2} u(x, y) \right) - 2x \left(\frac{\partial^2}{\partial y \partial x} u(x, y) \right) + \left(\frac{\partial^2}{\partial y^2} u(x, y) \right) = 0$$

Характеристики:

$$\{y = -\ln(x) + _C1\}$$

параболический тип

Замена переменных:

$$\{\eta = y, \xi = y + \ln(x)\}$$

Каноническая форма:

$$\left(\frac{\partial^2}{\partial \eta^2} u(\eta, \xi) \right) - \left(\frac{\partial}{\partial \xi} u(\eta, \xi) \right) = 0$$

Таким образом, каноническая форма уравнения будет

$$\frac{\partial^2 u}{\partial \eta^2} - \frac{\partial u}{\partial \xi} = 0.$$

Тот же пример, но другая замена переменных (для этого мы меняем в указанном месте `eta=y` на `eta=x`, после чего нажимаем клавишу <Enter> ввода программы `canonical_form` и затем снова применяем эту программу к решению задачи)

```
> canonical_form(equ);
```

Характеристики:

$$\{y = -\ln(x) + _C1\}$$

параболический тип

Замена переменных:

$$\{\eta = x, \xi = y + \ln(x)\}$$

Каноническая форма:

$$\left(\frac{\partial^2}{\partial \eta^2} u(\eta, \xi) \right) - \frac{\frac{\partial}{\partial \xi} u(\eta, \xi)}{\eta^2} = 0$$

Таким образом, другая каноническая форма уравнения

$$\frac{\partial^2 u}{\partial \eta^2} - \frac{1}{\eta^2} \frac{\partial u}{\partial \xi} = 0.$$

Очевидно, эта форма менее предпочтительна, так как уравнение получилось с переменными коэффициентами.

Задача 3.7. В каждой области, где сохраняется тип уравнения, привести к каноническому виду уравнение

$$y \frac{\partial^2 u}{\partial x^2} - x \frac{\partial^2 u}{\partial y^2} = 0.$$

Решение. Для решения задачи воспользуемся программой `canonical_form`.
Задаем уравнение

```
> a := y, 0, -x, 0, 0;
```

```
      a := y, 0, -x, 0, 0
```

```
> equ := a[1]*diff(u(x,y), x, x) + a[2]*diff(u(x,y), x, y) +
```

```
      a[3]*diff(u(x,y), y, y) + a[4]*diff(u(x,y), x) +
```

```
      a[5]*diff(u(x,y), y) = 0;
```

$$equ := y \left(\frac{\partial^2}{\partial x^2} u(x, y) \right) - x \left(\frac{\partial^2}{\partial y^2} u(x, y) \right) = 0$$

Рассматриваем уравнение в области $x < 0, y < 0$:

```
> assume(x<0, y<0); canonical_form(equ);
```

Характеристики:

$$\left\{ -\frac{2(-x)^{(3/2)}}{3} + \frac{2(-y)^{(3/2)}}{3} + _CI = 0, -\frac{2(-x)^{(3/2)}}{3} - \frac{2(-y)^{(3/2)}}{3} + _CI = 0 \right\}$$

гиперболический тип

Замена переменных:

$$\left\{ \xi = -\frac{2x\sqrt{-x}}{3}, \eta = -\frac{2y\sqrt{-y}}{3} \right\}$$

Каноническая форма:

$$\left(\frac{\partial^2}{\partial \eta^2} u(\eta, \xi) \right) - \left(\frac{\partial^2}{\partial \xi^2} u(\eta, \xi) \right) - \frac{1}{3} \frac{\left(\frac{\partial}{\partial \xi} u(\eta, \xi) \right) \eta - \left(\frac{\partial}{\partial \eta} u(\eta, \xi) \right) \xi}{\eta \xi} = 0$$

Таким образом, в области $x < 0, y < 0$ уравнение гиперболического типа и имеет канонический вид

$$\frac{\partial^2 u}{\partial \eta^2} - \frac{\partial^2 u}{\partial \xi^2} - \frac{\eta \frac{\partial u}{\partial \xi} - \xi \frac{\partial u}{\partial \eta}}{3\eta\xi} = 0.$$

Рассматриваем уравнение в области $x > 0, y > 0$:

> assume (x>0, y>0); canonical_form(equ);

Характеристики:

$$\{y^{(3/2)} + x^{(3/2)} - _CI = 0, y^{(3/2)} - x^{(3/2)} - _CI = 0\}$$

гиперболический тип

Замена переменных:

$$\{\xi = y^{(3/2)}, \eta = -x^{(3/2)}\}$$

Каноническая форма:

$$\left(\frac{\partial^2}{\partial \eta^2} u(\eta, \xi)\right) - \left(\frac{\partial^2}{\partial \xi^2} u(\eta, \xi)\right) - \frac{1}{3} \frac{\left(\frac{\partial}{\partial \xi} u(\eta, \xi)\right) \eta - \left(\frac{\partial}{\partial \eta} u(\eta, \xi)\right) \xi}{\eta \xi} = 0$$

То есть и в области $x > 0, y > 0$ уравнение гиперболического типа.

Рассматриваем уравнение в области $x > 0, y < 0$:

> assume (x>0, y<0); canonical_form(equ);

Характеристики:

$$\left\{ \frac{2(-y)^{(3/2)}}{3} - \frac{2}{3} I x^{(3/2)} = _CI, \frac{2(-y)^{(3/2)}}{3} + \frac{2}{3} I x^{(3/2)} = _CI \right\}$$

эллиптический тип

Замена переменных:

$$\left\{ \xi = \frac{2x^{(3/2)}}{3}, \eta = -\frac{2y\sqrt{-y}}{3} \right\}$$

Каноническая форма:

$$\left(\frac{\partial^2}{\partial \eta^2} u(\eta, \xi)\right) + \left(\frac{\partial^2}{\partial \xi^2} u(\eta, \xi)\right) + \frac{1}{3} \frac{\left(\frac{\partial}{\partial \xi} u(\eta, \xi)\right) \eta + \left(\frac{\partial}{\partial \eta} u(\eta, \xi)\right) \xi}{\xi \eta} = 0$$

Рассматриваем уравнение в области $x < 0, y > 0$:

> assume (x<0, y>0); canonical_form(equ);

Характеристики:

$$\left\{ \frac{2 y^{\sim(3/2)}}{3} - \frac{2}{3} I(-x^{\sim})^{(3/2)} = -CI, \frac{2 y^{\sim(3/2)}}{3} + \frac{2}{3} I(-x^{\sim})^{(3/2)} = -CI \right\}$$

эллиптический тип

Замена переменных:

$$\left\{ \eta = \frac{2 y^{\sim(3/2)}}{3}, \xi = -\frac{2 x^{\sim} \sqrt{-x^{\sim}}}{3} \right\}$$

Каноническая форма:

$$\left(\frac{\partial^2}{\partial \eta^2} u(\eta, \xi) \right) + \left(\frac{\partial^2}{\partial \xi^2} u(\eta, \xi) \right) + \frac{1}{3} \frac{\left(\frac{\partial}{\partial \xi} u(\eta, \xi) \right) \eta + \left(\frac{\partial}{\partial \eta} u(\eta, \xi) \right) \xi}{\xi \eta} = 0$$

Итак, в областях $x > 0, y < 0$ и $x < 0, y > 0$ уравнение эллиптического типа и имеет канонический вид

$$\frac{\partial^2 u}{\partial \eta^2} + \frac{\partial^2 u}{\partial \xi^2} + \frac{\eta \frac{\partial u}{\partial \xi} + \xi \frac{\partial u}{\partial \eta}}{3 \eta \xi} = 0.$$

Задача 3.8. Привести к каноническому виду уравнение

$$x^2 \frac{\partial^2 u}{\partial x^2} - y^2 \frac{\partial^2 u}{\partial y^2} = 0.$$

Решение. Для решения задачи воспользуемся программой `canonical_form`.
Задаем уравнение

```
> a:= x^2, 0, -y^2, 0, 0;
eq:=a[1]*diff(u(x,y),x,x)+a[2]*diff(u(x,y),x,y)+
a[3]*diff(u(x,y),y,y)+a[4]*diff(u(x,y),x)+
a[5]*diff(u(x,y),y)=0;
canonical_form(eq);
```

$$a := x^2, 0, -y^2, 0, 0$$

$$eq := x^2 \left(\frac{\partial^2}{\partial x^2} u(x, y) \right) - y^2 \left(\frac{\partial^2}{\partial y^2} u(x, y) \right) = 0$$

Характеристики:

$$\left\{ y = \frac{CI}{x}, y = -CI x \right\}$$

гиперболический тип

Замена переменных:

$$\left\{ \xi = \frac{y(1+x^2)}{2x}, \eta = -\frac{y(-1+x^2)}{2x} \right\}$$

Каноническая форма:

$$\left(\frac{\partial^2}{\partial \eta^2} u(\eta, \xi) \right) - \left(\frac{\partial^2}{\partial \xi^2} u(\eta, \xi) \right) + \frac{\left(\frac{\partial}{\partial \xi} u(\eta, \xi) \right) + \left(\frac{\partial}{\partial \eta} u(\eta, \xi) \right)}{-\eta + \xi} = 0$$

Иногда бывает удобнее другая форма канонического уравнения

> eq_can;

$$\left(\frac{\partial^2}{\partial \eta^2} u(\eta, \xi) \right) - \left(\frac{\partial^2}{\partial \xi^2} u(\eta, \xi) \right) + \frac{\left(\frac{\partial}{\partial \xi} u(\eta, \xi) \right) + \left(\frac{\partial}{\partial \eta} u(\eta, \xi) \right)}{-\eta + \xi} = 0$$

> tr:={xi=t+tau, eta=t-tau};

itr:={t=(xi+eta)/2, tau=(xi-eta)/2};

$$tr := \{ \xi = t + \tau, \eta = t - \tau \}$$

$$itr := \left\{ t = \frac{\eta}{2} + \frac{\xi}{2}, \tau = -\frac{\eta}{2} + \frac{\xi}{2} \right\}$$

> PDEtools[dchange](tr, lhs(eq_can), itr, [t, tau], simplify)=0;

$$\frac{1}{2} \frac{2 \left(\frac{\partial^2}{\partial \tau \partial t} u(t, \tau) \right) \tau - \left(\frac{\partial}{\partial t} u(t, \tau) \right)}{\tau} = 0$$

Последнее уравнение легко интегрируется (как в задаче 3.1). Впрочем, общие решения этих канонических уравнений можно найти и с помощью Maple

> pdsolve(%);

$$u(t, \tau) = _F1(\tau) + _F2(t) \sqrt{\tau}$$

> pdsolve(eq_can);

$$u(\eta, \xi) = _F1(-\eta + \xi) + _F2(\eta + \xi) \sqrt{-\eta + \xi}$$

Здесь мы воспользовались еще одной командой Maple — функцией pdsolve. Кратко прокомментируем эту команду. Функция pdsolve предназначена для исследования уравнений в частных производных и находится в основной библиотеке Maple. Кроме того, для исследования уравнений в частных производных в Maple имеется специальный набор функций пакета PDEtools (одной такой функцией dchange из этого пакета мы уже пользовались). В некотором смысле функция pdsolve аналогична функции dsolve.

Функция pdsolve может использоваться в одном из следующих форматов:

pdsolve (PDE, f, HINT, INTEGRATE, build)

pdsolve (PDE_system, funcs, HINT, other_options)

```
pdsolve (PDE_system, conds, numeric, other_options)
pdsolve (PDE_system, conds, type=numeric, other_options)
```

Параметрами функции `pdsolve` являются:

- `PDE` — дифференциальное уравнение в частных производных;
- `PDE_system` — система дифференциальных уравнений в частных производных, может содержать неравенства;
- `conds` — начальные или граничные условия;
- `f` — неизвестная функция или имя (этот параметр требуется для указания функции, относительно которой рассматривается уравнение, когда в составе уравнения имеются несколько функций);
- `funcs` — (необязательный) множество или список с неизвестными функциями или именами;
- `HINT` — (необязательный) уравнение вида `HINT = argument`, где `argument` — одно из следующих допустимых значений: '+', '*', любое алгебраическое выражение, или 'strip'. Этот параметр представляет собой входную пользовательскую подсказку, указывающую, в какой форме искать решение — разделение переменных в виде суммы, произведения или в какой-либо другой форме; последнее допустимое значение этой опции 'strip' применяется к уравнениям первого порядка и указывает на необходимость использования метода характеристик для получения общего решения;
- `INTEGRATE` — (необязательный), указывает на необходимость автоматического интегрирования множества обыкновенных дифференциальных уравнений, когда уравнение в частных производных решается с помощью разделения переменных;
- `build` — (необязательный) указывает на необходимость сформировать явное выражение для "неизвестной функции";
- `numeric` — ключевое слово, указывающее, что должно быть найдено численное решение;
- `other_options` — при точном решении системы уравнений, все опции, принимаемые функцией `casesplit` также принимаются функцией `pdsolve`.

Необязательные параметры можно задавать по одному или в связке в любом порядке.

Для первых двух вариантов вызова функции `pdsolve`, когда задано одно уравнение или система уравнений, возможно включая обыкновенные дифференциальные уравнения, алгебраические ограничения и неравенства, главная задача `pdsolve` заключается в том, чтобы найти аналитическое решение. При этом не накладывается никаких ограничений относительно типа уравнения, порядка уравнения, или числа зависимых или независимых переменных уравнения в частных производных или системы таких уравнений.

Остальные два варианта вызова функции `pdsolve` предназначены для нахождения численного решения заданных на входе уравнений в частных производных или систем. Описание использования `pdsolve` в этих случаях, типы уравнений в

частных производных, которые могут быть численно решены, и другие параметры, мы рассматриваем в специальной главе, посвященной численным методам. Любая справка по этому вопросу, как обычно, может быть получена с помощью вызова `?pdsolve, numeric`.

Функция `pdsolve` к настоящему моменту времени распознает некоторое количество семейств уравнений в частных производных, которые могут быть решены, используя стандартные методы. Когда данное уравнение в частных производных принадлежит непризнанному семейству, `pdsolve` использует эвристический алгоритм, который делает попытку разделения переменных, основанного на специфической структуре уравнения в частных производных.

Стратегия использования `pdsolve` состоит в том, чтобы найти наиболее общее решение данного уравнения в частных производных или, в самом худшем случае, пытаться искать полное разделение переменных. Таким образом, когда эти задачи завершаются успешно, функция `pdsolve` возвращает результат в одном из следующих видов:

- общее решение;
- так называемое квази-общее решение — решение, содержащее произвольные функции, но не в достаточном количестве или не имеющее достаточно переменных, чтобы составить общее решение;
- набор отдельных обыкновенных дифференциальных уравнений относительно всех разделенных переменных, или завершённое решение, полученное после интегрирования этого набора (когда задана опция `INTEGRATE`).

В первом случае решение представляется в виде, содержащем произвольные функции `_Fn`, которые предполагаются достаточно гладкими функциями своих параметров. Их количество позволяет удовлетворить любым допустимым дополнительным условиям (начальным и/или граничным).

Во втором случае количество произвольных функций не может обеспечить решение задачи с заданными дополнительными условиями. Функция `pdsolve` выражает результат, используя для отображения решения специальную внутреннюю структуру `PDESolStruc`. В этой структуре перечисляются разделенные переменные, и после ключевого слова `&where` отображается набор обыкновенных дифференциальных уравнений, которым удовлетворяют разделенные переменные. Структура `PDESolStruc` показывает, как формируется частное решение.

Любые произвольные постоянные, вводимые при отделении переменных представлены как переменные с индексами `_c[1]`, `_c[2]`, ..., и являются глобальными переменными. Этими возможностями функции `pdsolve` мы будем пользоваться в главе 4, когда будем решать задачи методом Фурье разделения переменных.

Вернемся теперь к нашей задаче. В результате работы команды `pdsolve` мы получили общее решение заданного уравнения в следующей форме

$$u(\xi, \eta) = F_1(\xi - \eta) + F_2(\xi + \eta)\sqrt{\xi - \eta},$$

где $F_1(t)$ и $F_2(t)$ — любые достаточно гладкие функции, которые в Maple обозначены как `_F1` и `_F2`.

Задача 3.9. Найти общее решение уравнения с постоянными коэффициентами

$$\frac{\partial^2 u}{\partial x^2} - 2 \frac{\partial u}{\partial x \partial y} - 3 \frac{\partial^2 u}{\partial y^2} = 0.$$

Решение. Для решения задачи воспользуемся стандартными средствами Maple. Зададим уравнение

```
> eq := diff(u(x, y), x, x) - 2*diff(u(x, y), x, y) -
3*diff(u(x, y), y, y) = 0;
```

$$eq := \left(\frac{\partial^2}{\partial x^2} u(x, y) \right) - 2 \left(\frac{\partial^2}{\partial y \partial x} u(x, y) \right) - 3 \left(\frac{\partial^2}{\partial y^2} u(x, y) \right) = 0$$

Воспользуемся стандартной программой Maple — `mapde(eq, canom)`. Эта программа преобразовывает исходное уравнение в возможно более простое, которое можно решить. Фактически, это еще один способ приведения заданного уравнения к каноническому виду. Разработчики системы, правда, предупреждают, что команда `mapde` всегда будет успешно выполнена, когда коэффициенты уравнения постоянны. С переменными коэффициентами команда успешно выполняется только иногда. Для того чтобы воспользоваться этой командой, надо подключить пакет `PDEtools`

```
> with(PDEtools);
```

```
> mapde(eq, canom);
```

$$\sqrt{16} \left(\frac{\partial^2}{\partial_{\xi 2} \partial_{\xi 1}} u(\xi 1, \xi 2) \right) \&where \{ \xi 1 = 3x + y, \xi 2 = \frac{x}{4} - \frac{y}{4} \}$$

```
> op(%);
```

$$\sqrt{16} \left(\frac{\partial^2}{\partial_{\xi 2} \partial_{\xi 1}} u(\xi 1, \xi 2) \right), \{ \xi 1 = 3x + y, \xi 2 = \frac{x}{4} - \frac{y}{4} \}$$

Найдем теперь общее решение полученного уравнения гиперболического типа с помощью команды `pdsolve`

```
> pdsolve(%[1]);
```

$$u(\xi 1, \xi 2) = _F2(\xi 1) + _F1(\xi 2)$$

Вернемся к старым переменным

```
> sol := u(x, y) = subs(%%[2], rhs(%));
```

$$sol := u(x, y) = _F2(3x + y) + _F1\left(\frac{x}{4} - \frac{y}{4}\right)$$

Проверим найденное решение

```
> simplify(subs(sol, eq));
```

$$\left(\frac{\partial^2}{\partial x^2}\left(-F_2(3x+y) + F_1\left(\frac{x}{4} - \frac{y}{4}\right)\right)\right) - 2\left(\frac{\partial^2}{\partial y \partial x}\left(-F_2(3x+y) + F_1\left(\frac{x}{4} - \frac{y}{4}\right)\right)\right) - 3\left(\frac{\partial^2}{\partial y^2}\left(-F_2(3x+y) + F_1\left(\frac{x}{4} - \frac{y}{4}\right)\right)\right) = 0$$

> simplify(lhs(%));

0

Итак, общее решение имеет вид

$$u(x, y) = F_1\left(\frac{x}{4} - \frac{y}{4}\right) + F_2(3x + y),$$

где $F_1(t)$ и $F_2(t)$ — любые достаточно гладкие функции

Задача 3.10. Найти общее решение уравнения с постоянными коэффициентами

$$3\frac{\partial^2 u}{\partial x^2} - 5\frac{\partial u}{\partial x \partial y} - 2\frac{\partial^2 u}{\partial y^2} + 3\frac{\partial u}{\partial x} + \frac{\partial u}{\partial y} = 2.$$

Решение. Зададим уравнение

> eq := 3*diff(u(x, y), x, x) - 5*diff(u(x, y), x, y) -
2*diff(u(x, y), y, y) + 3*diff(u(x, y), x) + diff(u(x, y), y) = 2;

$$eq := 3\left(\frac{\partial^2}{\partial x^2} u(x, y)\right) - 5\left(\frac{\partial^2}{\partial y \partial x} u(x, y)\right) - 2\left(\frac{\partial^2}{\partial y^2} u(x, y)\right) + 3\left(\frac{\partial}{\partial x} u(x, y)\right) + \left(\frac{\partial}{\partial y} u(x, y)\right) = 2$$

Преобразуем к канонической форме

> mapde(eq, canom);

$$\left(\frac{1}{98}(3\sqrt{49} - 21)\sqrt{49}\left(\frac{\partial}{\partial \xi_2} u(\xi_1, \xi_2)\right) + \frac{1}{6}(3\sqrt{49} + 21)\left(\frac{\partial}{\partial \xi_1} u(\xi_1, \xi_2)\right) - 2 + \sqrt{49}\left(\frac{\partial^2}{\partial \xi_2 \partial \xi_1} u(\xi_1, \xi_2)\right)\right) \&where \{ \xi_1 = 2x + y, \xi_2 = \frac{x}{7} - \frac{3y}{7} \}$$

> op(%);

$$\frac{1}{98}(3\sqrt{49} - 21)\sqrt{49}\left(\frac{\partial}{\partial \xi_2} u(\xi_1, \xi_2)\right) + \frac{1}{6}(3\sqrt{49} + 21)\left(\frac{\partial}{\partial \xi_1} u(\xi_1, \xi_2)\right) - 2 + \sqrt{49}\left(\frac{\partial^2}{\partial \xi_2 \partial \xi_1} u(\xi_1, \xi_2)\right), \{ \xi_1 = 2x + y, \xi_2 = \frac{x}{7} - \frac{3y}{7} \}$$

Решим уравнение

> pdsolve(%[1]);

$$u(\xi_1, \xi_2) = F_1(\xi_2) + e^{(-\xi_2)} F_2(\xi_1) + \frac{2\xi_1}{7}$$

> sol := u(x, y) = subs(%%[2], rhs(%));

$$\text{sol} := u(x, y) = _F1\left(\frac{x}{7} - \frac{3y}{7}\right) + e^{\left(-\frac{x}{7} + \frac{3y}{7}\right)} _F2(2x + y) + \frac{4x}{7} + \frac{2y}{7}$$

Проверим полученное решение

```
> simplify(subs(sol, eq));
```

$$3 \left(\frac{\partial^2}{\partial x^2} \left(_F1\left(\frac{x}{7} - \frac{3y}{7}\right) + e^{\left(-\frac{x}{7} + \frac{3y}{7}\right)} _F2(2x + y) + \frac{4x}{7} + \frac{2y}{7} \right) \right)$$

$$- 5 \left(\frac{\partial^2}{\partial y \partial x} \left(_F1\left(\frac{x}{7} - \frac{3y}{7}\right) + e^{\left(-\frac{x}{7} + \frac{3y}{7}\right)} _F2(2x + y) + \frac{4x}{7} + \frac{2y}{7} \right) \right)$$

$$- 2 \left(\frac{\partial^2}{\partial y^2} \left(_F1\left(\frac{x}{7} - \frac{3y}{7}\right) + e^{\left(-\frac{x}{7} + \frac{3y}{7}\right)} _F2(2x + y) + \frac{4x}{7} + \frac{2y}{7} \right) \right)$$

$$+ 3 \left(\frac{\partial}{\partial x} \left(_F1\left(\frac{x}{7} - \frac{3y}{7}\right) + e^{\left(-\frac{x}{7} + \frac{3y}{7}\right)} _F2(2x + y) + \frac{4x}{7} + \frac{2y}{7} \right) \right)$$

$$+ \left(\frac{\partial}{\partial y} \left(_F1\left(\frac{x}{7} - \frac{3y}{7}\right) + e^{\left(-\frac{x}{7} + \frac{3y}{7}\right)} _F2(2x + y) + \frac{4x}{7} + \frac{2y}{7} \right) \right) = 2$$

```
> simplify(lhs(%));
```

2

Итак, общее решение имеет вид

$$u(x, y) = \frac{4x}{7} + \frac{2y}{7} + F_1\left(\frac{x}{7} - \frac{3y}{7}\right) + e^{-\frac{x}{7} + \frac{3y}{7}} F_2(2x + y),$$

где $F_1(t)$ и $F_2(t)$ — любые достаточно гладкие функции.

Отметим, что общее решение исходного заданного уравнения можно найти сразу с помощью команды `pdsolve`

```
> pdsolve(eq);
```

$$u(x, y) = _F1(3y - x) + e^{(-x)} _F2(y + 2x) + \frac{2x}{3} - \frac{2}{3}$$

Таким образом, общее решение можно представить и в такой форме

$$u(x, y) = \frac{2x}{3} - \frac{2}{3} + F_1(3y - x) + e^{-x} F_2(2x + y).$$

Задача 3.11. Найти общее решение уравнения с постоянными коэффициентами

$$\frac{\partial u}{\partial x \partial y} - 2 \frac{\partial u}{\partial x} - 3 \frac{\partial u}{\partial y} + 6u = 2e^{x+y}.$$

Решение. Зададим уравнение

> eq:=diff(u(x,y),x,y)-2*diff(u(x,y),x)-
3*diff(u(x,y),y)+6*u(x,y)=2*exp(x+y);

$$eq := \left(\frac{\partial^2}{\partial y \partial x} u(x, y) \right) - 2 \left(\frac{\partial}{\partial x} u(x, y) \right) - 3 \left(\frac{\partial}{\partial y} u(x, y) \right) + 6 u(x, y) = 2 e^{(x+y)}$$

> mapde(eq, canom);

$$\left(\left(\frac{\partial^2}{\partial x \partial y} u(x, y) \right) - 2 \left(\frac{\partial}{\partial x} u(x, y) \right) - 3 \left(\frac{\partial}{\partial y} u(x, y) \right) + 6 u(x, y) - 2 e^{(x+y)} \right) \& \text{where}$$

the received pde is already in canom form

Система Maple сообщает нам, что уравнение уже имеет канонический вид (какого типа это уравнение?). Решим это уравнение

> sol:=pdsolve(eq);

$$sol := u(x, y) = e^{(2y)}_F1(x) + e^{(3x)}_F2(y) + e^{(x+y)}$$

Выполним проверку найденного решения

> simplify(subs(sol, eq));

$$\left(\frac{\partial^2}{\partial y \partial x} (e^{(2y)}_F1(x) + e^{(3x)}_F2(y) + e^{(x+y)}) \right) - 2 \left(\frac{\partial}{\partial x} (e^{(2y)}_F1(x) + e^{(3x)}_F2(y) + e^{(x+y)}) \right) - 3 \left(\frac{\partial}{\partial y} (e^{(2y)}_F1(x) + e^{(3x)}_F2(y) + e^{(x+y)}) \right) + 6 e^{(2y)}_F1(x) + 6 e^{(3x)}_F2(y) + 6 e^{(x+y)} = 2 e^{(x+y)}$$

> simplify(lhs(%));

$$2 e^{(x+y)}$$

Таким образом, общее решение уравнения

$$u(x, y) = e^{2y} F_1(x) + e^{3x} F_2(y) + e^{x+y}.$$

Задача 3.12. Решить задачу о колебаниях струны $0 < x < l$ с закрепленными концами, если начальные скорости точек струны равны нулю, а начальное отклонение имеет форму синусоиды

$$u|_{t=0} = A \sin\left(\frac{\pi x}{l}\right).$$

Решение. Эта задача может быть решена методом Фурье (см. главу 4). Можно, однако, использовать для решения данной задачи и формулу Даламбера. Покажем, как это сделать.

Нам необходимо проинтегрировать уравнение колебаний струны при начальных данных

$$u|_{t=0} = \varphi(x) = A \sin\left(\frac{\pi x}{l}\right), \quad \frac{\partial u}{\partial t}|_{t=0} = \psi(x) = 0,$$

и при краевых условиях

$$u|_{x=0} = 0, \quad u|_{x=l} = 0.$$

Продолжим функции $\varphi(x)$ и $\psi(x)$ на отрезок $[-l, 0]$ нечетным образом, а затем периодически на всю ось. В результате получим функции

$$\Phi(x) = A \sin\left(\frac{\pi x}{l}\right), \quad \Psi(x) = 0, \quad -\infty < x < \infty,$$

совпадающие на отрезке $[0, l]$ с функциями $\varphi(x)$ и $\psi(x)$.

Подставляя функции $\Phi(x)$ и $\Psi(x)$ в формулу Даламбера (3.17), находим решение уравнения колебаний в виде

$$u(x, t) = \frac{1}{2} \left\{ A \sin\left[\frac{\pi}{l}(x - at)\right] + A \sin\left[\frac{\pi}{l}(x + at)\right] \right\},$$

или

$$u(x, t) = A \sin\left(\frac{\pi x}{l}\right) \cos\left(\frac{\pi at}{l}\right).$$

Задача 3.13. Решить следующую задачу Коши для одномерного волнового уравнения

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + e^x, \quad u|_{t=0} = \sin x, \quad \frac{\partial u}{\partial t}|_{t=0} = x + \cos x.$$

Решение. Воспользуемся обобщением формулы Даламбера (3.21) на случай неоднородного волнового уравнения. Определим искомую функцию в Maple в соответствии с этой формулой:

```
> unprotect (Psi);
u := (x, t) -> (Phi (x+v*t) + Phi (x-v*t)) / 2 +
int (Psi (xi), xi=x-v*t..x+v*t) / (2*v) +
int (int (f (xi, tau), xi=x-v*(t-tau)..x+v*(t-tau)),
tau=0..t) / (2*v);
```

$$u := (x, t) \rightarrow \frac{1}{2} \Phi(x + vt) + \frac{1}{2} \Phi(x - vt) + \frac{1}{2} \left(\frac{1}{v} \int_{x-vt}^{x+vt} \Psi(\xi) d\xi \right) + \frac{1}{2} \left(\frac{1}{v} \int_0^t \int_{x-v(t-\tau)}^{x+v(t-\tau)} f(\xi, \tau) d\xi d\tau \right)$$

Здесь первая команда, `unprotect (Psi)`, просто снимает защиту с символьной переменной `Psi` («по умолчанию» эта переменная обозначает логарифмическую производную от гамма-функции — пси-функцию). Чтобы сохранить за переменной `Psi` смысл начальной скорости мы и выполнили предварительно эту команду.

Теперь зададим коэффициент волнового уравнения $v^2 = 1$, начальные функции $\Phi(x) = \sin x$, $\Psi(x) = x + \cos x$ и правую часть $f(x, t) = e^x$. Таким образом, мы получаем

```
> v:=1;Phi:=x->sin(x);Psi:=x->x+cos(x);
f:=(x,t)->exp(x);u(x,t);
v:=1
Phi:=sin
Psi:=x->x+cos(x)
f:=(x,t)->e^x
sin(x+t)+xt-e^x+1/2*e^(x+t)+1/2*e^(x-t)
```

Итак, решение нашей задачи имеет вид

$$u = xt + \sin(x+t) + \frac{e^{x+t} + e^{x-t}}{2} - e^x.$$

Очевидно, экспоненты можно преобразовать к гиперболическим функциям. Сделаем это тоже с помощью Maple. Для начала выделим в отдельное выражение эти экспоненты и вынесем за скобки общий множитель e^x

```
> expand(op(3,%%)+op(4,%%)+op(5,%%));collect(%,exp);
-e^x+1/2*e^x*e^t+1/2*e^x/e^t
(-1+1/2*e^t+1/2*1/e^t)*e^x
```

Теперь упростим выражение в скобках

```
> simplify(convert(op(1,%),trig));
-1+cosh(t)
```

Итак, окончательно получаем

$$u = xt + \sin(x+t) + e^x(\operatorname{ch} t - 1).$$

Задача 3.14. Решить следующую задачу Коши для одномерного волнового уравнения

$$\frac{\partial^2 u}{\partial t^2} = 4 \frac{\partial^2 u}{\partial x^2} + xt, \quad u|_{t=0} = x^2, \quad \frac{\partial u}{\partial t}|_{t=0} = x.$$

Решение. Воспользуемся уже определенной в задаче 3.13 функцией $u(x, t)$. Зададим коэффициент волнового уравнения, начальные функции и правую часть в Maple. Получим

```
> v:=2;Phi:=x->x^2;Psi:=x->x;f:=(x,t)->x*t;u(x,t);
```

$$v := 2$$

$$\Phi := x \rightarrow x^2$$

$$\Psi := x \rightarrow x$$

$$f := (x, t) \rightarrow x t$$

$$\frac{5(x+2t)^2}{8} + \frac{3(x-2t)^2}{8} - \frac{x t^3}{3} + \frac{\left(\frac{(x+2t)^2}{2} - \frac{(x-2t)^2}{2}\right) t^2}{8}$$

Упростим полученное решение

```
> simplify(%);
```

$$x^2 + x t + 4 t^2 + \frac{1}{6} x t^3$$

Итак, $u = x^2 + x t + 4 t^2 + \frac{1}{6} x t^3$.

Задача 3.15. Решить следующую задачу Коши для двумерного волнового уравнения

$$\frac{\partial^2 u}{\partial t^2} = v^2 \left(\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} \right), \quad u|_{t=0} = 0, \quad \frac{\partial u}{\partial t} \Big|_{t=0} = (x-y)^2.$$

Решение. Воспользуемся формулой Пуассона (3.22) для функции $u(x, y, t)$. Определим задачу в Maple — подключим необходимые пакеты и введем уравнение и дополнительные условия:

```
> restart;with(linalg):with(student):
```

```
Warning, the protected names norm and trace have been
redefined and unprotected
```

```
unprotect(Psi):
```

```
eq:=diff(u(x,y,t),t,t)=v^2*laplacian(u(x,y,t),[x,y]);
```

```
ic:=u(x,y,0)=Phi(x,y);ict:=D[3](u)(x,y,0)=Psi(x,y);
```

$$eq := \frac{\partial^2}{\partial t^2} u(x, y, t) = v^2 \left(\left(\frac{\partial^2}{\partial x^2} u(x, y, t) \right) + \left(\frac{\partial^2}{\partial y^2} u(x, y, t) \right) \right)$$

$$ic := u(x, y, 0) = \Phi(x, y)$$

$$ict := D_3(u)(x, y, 0) = \Psi(x, y)$$

```
> assume(v>0,t>0);
```

Для записи дифференциального уравнения мы воспользовались оператором — лапласианом `laplacian`, который находится в пакете `linalg`. Команда может вызываться в двух вариантах:

- `laplacian(f,v);`

- `laplacian(f, v, co)`;

Здесь f — скалярное выражение; v — вектор или список переменных; co — используется для вычисления лапласиана в ортогональных криволинейных системах.

Определим формулу Пуассона в Maple:

```
> 1/(2*Pi*v)*Doubleint(Psi(xi)/
sqrt((v*t)^2-r^2),xi[1],xi[2],S)
+1/(2*Pi*v)*Diff(Doubleint(Phi(xi)/
sqrt((v*t)^2-r^2),xi[1],xi[2],S),t);
```

$$\frac{1}{2} \left(\frac{1}{\pi v} \iint_S \frac{\Psi(\xi)}{\sqrt{v^2 t^2 - r^2}} d\xi_1 d\xi_2 \right) + \frac{1}{2} \frac{\frac{\partial}{\partial t} \left(\iint_S \frac{\Phi(\xi)}{\sqrt{v^2 t^2 - r^2}} d\xi_1 d\xi_2 \right)}{\pi v}$$

Здесь область интегрирования в двойном интеграле

$$S = \{(\xi_1, \xi_2) : (\xi_1 - x)^2 + (\xi_2 - y)^2 \leq v^2 t^2\}.$$

Вводим начальные условия

```
> Phi := (x, y) -> 0; Psi := (x, y) -> (x-y)^2;
Phi := 0
Psi := (x, y) -> (x-y)^2
```

Для вычисления интегралов используем полярные координаты:

```
> T := {xi=x+r*cos(phi), eta=y+r*sin(phi)};
T := {xi = x + r cos(phi), eta = y + r sin(phi)}
```

```
> subs(T, 1/(2*Pi*v)*
Doubleint(Psi(xi,eta)/sqrt((v*t)^2-r^2)*r,
phi=0..2*Pi,r=0..v*t)+1/(2*Pi*v)*Diff(
Doubleint(Phi(xi,eta)/sqrt((v*t)^2-r^2)*r,
phi=0..2*Pi,r=0..v*t),t));
```

$$\frac{1}{2} \left(\frac{1}{\pi v} \int_0^{v \cdot t} \int_0^{2\pi} \frac{(x + r \cos(\phi) - y - r \sin(\phi))^2 r}{\sqrt{v^2 t^2 - r^2}} d\phi dr \right) + \frac{1}{2} \frac{d}{dt} \left(\int_0^{v \cdot t} \int_0^{2\pi} 0 d\phi dr \right)$$

Вычисляем интегралы:

```
> sol:=simplify(value(%));
```

$$sol := \frac{t (2 v^2 t^2 + 3 x^2 - 6 x y + 3 y^2)}{3}$$

```
> completesquare(%, x);
```

$$(x - y)^2 t + \frac{2 v^2 t^3}{3}$$

Здесь мы воспользовались еще одной полезной командой — `completesquare`. Эта команда-функция имеет два формата:

- `completesquare(f)`;
- `completesquare(f, x)`;

Здесь f — алгебраическое выражение; x — одна или более неизвестных, входящих в f . Эта команда находится в пакете `student` и выполняет выделение полного квадрата для полиномов второй степени относительно заданной переменной (или списка переменных).

Выполним проверку решения:

```
> u:=unapply(%, (x, y, t)):
```

```
simplify(rhs(eq)-lhs(eq));
```

```
simplify(rhs(ict)-lhs(ict));
```

```
simplify(rhs(ict)-lhs(ict));
```

0

0

0

Итак, мы получили $u(x, y, t) = (x - y)^2 t + \frac{2v^2 t^3}{3}$.

Задача 3.16. Решить следующую задачу Коши для трехмерного волнового уравнения

$$\frac{\partial^2 u}{\partial t^2} = \frac{1}{3} \left(\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} + \frac{\partial^2 u}{\partial z^2} \right), \quad u|_{t=0} = x^2 + y^2 + z^2, \quad \frac{\partial u}{\partial t} \Big|_{t=0} = 0.$$

Решение. Воспользуемся формулой Кирхгофа (3.23) для функции $u(x, y, z, t)$.

Определим задачу в Maple

```
> restart;with(linalg):with(student):
```

```
Warning, the protected names norm and trace have been  
redefined and unprotected
```

```
unprotect(Psi):
```

```

eq:=diff (u (x, y, z, t) , t, t) =
v^2*laplacian (u (x, y, z, t) , [x, y, z] );
ic:=u (x, y, z, 0)=Phi; ict:=D[4] (u) (x, y, z, 0)=Psi;
eq :=  $\frac{\partial^2}{\partial t^2} u(x, y, z, t) =$ 

$$v^2 \left( \left( \frac{\partial^2}{\partial x^2} u(x, y, z, t) \right) + \left( \frac{\partial^2}{\partial y^2} u(x, y, z, t) \right) + \left( \frac{\partial^2}{\partial z^2} u(x, y, z, t) \right) \right)$$

ic := u(x, y, z, 0) =  $\Phi$ 
ict :=  $D_4(u)(x, y, z, 0) = \Psi$ 

```

Определим формулу Кирхгофа в Maple:

```

> spr:=1/(4*Pi*v^2*t)*Int (Psi (xi) , S) +
1/(4*Pi*v^2)*Diff (Int (Phi (xi) , S) /t, t);

$$spr := \frac{1}{4} \left( \frac{1}{\pi v^2 t} \int \Psi(\xi) dS \right) + \frac{1}{4} \frac{\frac{\partial}{\partial t} \left( \frac{1}{t} \int \Phi(\xi) dS \right)}{\pi v^2}$$


```

Здесь область интегрирования в двойном интеграле

$$S = \left\{ (\xi_1, \xi_2, \xi_3) : (\xi_1 - x)^2 + (\xi_2 - x)^2 + (\xi_3 - z)^2 = v^2 t^2 \right\}.$$

Вводим исходные данные

```

> v:=1/sqrt (3) ; Phi :=x^2+y^2+z^2; Psi:=0;

$$v := \frac{\sqrt{3}}{3}$$


$$\Phi := x^2 + y^2 + z^2$$


$$\Psi := 0$$


```

Задачу решаем в сферических координатах:

```

> T:={x=x+r*cos (phi) *sin (theta) ,
y=y+r*sin (phi) *sin (theta) ,
z=z+r*cos (theta) };
T := {x = x + r cos(φ) sin(θ), y = y + r sin(φ) sin(θ), z = z + r cos(θ)}
> subs (Int (Psi (xi) , S) =
Doubleint (subs (T, Psi) *r^2*sin (theta) ,
theta=0..Pi, phi=0..2*Pi) ,
Int (Phi (xi) , S) =
Doubleint (subs (T, Phi) *r^2*sin (theta) ,

```

theta=0..Pi, phi=0..2*Pi), r=v*t, spr);

$$\frac{3}{4} \left(\frac{1}{\pi t} \int_0^{2\pi} \int_0^\pi 0 \, d\theta \, d\phi \right) + \frac{3}{4} \frac{\partial}{\partial t} \left(\frac{1}{t} \int_0^{2\pi} \int_0^\pi \frac{1}{3} \left(\left(x + \frac{1}{3} \sqrt{3} t \cos(\phi) \sin(\theta) \right)^2 + \left(y + \frac{1}{3} \sqrt{3} t \sin(\phi) \sin(\theta) \right)^2 + \left(z + \frac{1}{3} \sqrt{3} t \cos(\theta) \right)^2 \right) t^2 \sin(\theta) \, d\theta \, d\phi \right) / \pi$$

Вычисляем интегралы:

```
> sol:=simplify(value(%));
      sol := x^2 + t^2 + y^2 + z^2
```

Выполним проверку решения:

```
> u:=unapply(%, (x, y, z, t));
      simplify(rhs(eq)-lhs(eq));
      simplify(rhs(ic)-lhs(ic));
      simplify(rhs(ict)-lhs(ict));
      0
      0
      0
```

Итак, $u(x, y, z, t) = x^2 + y^2 + z^2 + t^2$.

Задача 3.17. Найти решение системы

$$\begin{cases} \frac{\partial w}{\partial x} - \frac{1}{E} \frac{\partial \sigma}{\partial t} = 0, \\ \frac{\partial \sigma}{\partial x} - \rho \frac{\partial w}{\partial t} = 0, \quad E, \rho = \text{const}, \end{cases}$$

описывающей упругие продольные колебания стержня, отвечающие начальным данным $w(x, 0) = \varphi(x)$, $\sigma(x, 0) = \psi(x)$.

Решение. Определим заданную систему уравнений в Maple:

```
> eq1:=diff(w(x,t), x)-diff(sigma(x,t), t)/E=0;
      eq2:=diff(sigma(x,t), x)-rho*diff(w(x,t), t)=0;
```

$$\begin{aligned} eq1 &:= \left(\frac{\partial}{\partial x} w(x, t) \right) - \frac{\frac{\partial}{\partial t} \sigma(x, t)}{E} = 0 \\ eq2 &:= \left(\frac{\partial}{\partial x} \sigma(x, t) \right) - \rho \left(\frac{\partial}{\partial t} w(x, t) \right) = 0 \end{aligned}$$

Находим общее решение данной системы

```
> ans:=pdsolve({eq1,eq2},{w(x,t),sigma(x,t)});
```

$$ans := \left\{ \begin{array}{l} \sigma(x, t) = E \sqrt{\frac{\rho}{E}} {}_2F_2\left(\sqrt{\frac{\rho}{E}} x + t\right) - E \sqrt{\frac{\rho}{E}} {}_2F_1\left(\frac{x}{2} - \frac{t}{2\sqrt{\frac{\rho}{E}}}\right) + {}_2C1, \\ w(x, t) = {}_2F_2\left(\sqrt{\frac{\rho}{E}} x + t\right) + {}_2F_1\left(\frac{x}{2} - \frac{t}{2\sqrt{\frac{\rho}{E}}}\right) \end{array} \right\}$$

Выберем теперь произвольные функции ${}_2F_1$ и ${}_2F_2$ так, чтобы удовлетворялись начальные условия. Для этого определим найденные общие решения как функции от x и t :

```
> assign(ans):Sigma:=unapply(sigma(x,t),x,t);
```

```
W:=unapply(w(x,t),x,t);
```

$$\Sigma := (x, t) \rightarrow E \sqrt{\frac{\rho}{E}} {}_2F_2\left(\sqrt{\frac{\rho}{E}} x + t\right) - E \sqrt{\frac{\rho}{E}} {}_2F_1\left(\frac{1}{2}x - \frac{1}{2}\frac{t}{\sqrt{\frac{\rho}{E}}}\right) + {}_2C1$$

$$W := (x, t) \rightarrow {}_2F_2\left(\sqrt{\frac{\rho}{E}} x + t\right) + {}_2F_1\left(\frac{1}{2}x - \frac{1}{2}\frac{t}{\sqrt{\frac{\rho}{E}}}\right)$$

Подставим эти общие решения в начальные условия, получим

```
> e1:=Sigma(x,0)=psi(x);e2:=W(x,0)=phi(x);
```

$$e1 := E \sqrt{\frac{\rho}{E}} {}_2F_2\left(\sqrt{\frac{\rho}{E}} x\right) - E \sqrt{\frac{\rho}{E}} {}_2F_1\left(\frac{x}{2}\right) + {}_2C1 = \psi(x)$$

$$e2 := {}_2F_2\left(\sqrt{\frac{\rho}{E}} x\right) + {}_2F_1\left(\frac{x}{2}\right) = \phi(x)$$

Решим теперь систему относительно неизвестных функций

```
> solve({e1,e2},{_2F1(x/2),_2F2((1/E*rho)^(1/2)*x)});
```

$$\left\{ \begin{array}{l} {}_2F_2\left(\sqrt{\frac{\rho}{E}} x\right) = \frac{1}{2} \frac{E \sqrt{\frac{\rho}{E}} \phi(x) - {}_2C1 + \psi(x)}{E \sqrt{\frac{\rho}{E}}}, \\ {}_2F_1\left(\frac{x}{2}\right) = \frac{1}{2} \frac{E \sqrt{\frac{\rho}{E}} \phi(x) + {}_2C1 - \psi(x)}{E \sqrt{\frac{\rho}{E}}} \end{array} \right\}$$

Итак, мы получили

```
> assign(%):
```


> _F2 ((1/E*rho)^(1/2)*x):subs(x=x/(1/E*rho)^(1/2),%):
 _F2:=unapply(%,x);

$$_F2 := x \rightarrow \frac{1}{2} \frac{E \sqrt{\frac{\rho}{E}} \phi\left(\frac{x}{\sqrt{\frac{\rho}{E}}}\right) - _CI + \psi\left(\frac{x}{\sqrt{\frac{\rho}{E}}}\right)}{E \sqrt{\frac{\rho}{E}}}$$

> _F1(x/2):subs(x=2*x,%):_F1:=unapply(%,x);

$$_F1 := x \rightarrow \frac{1}{2} \frac{E \sqrt{\frac{\rho}{E}} \phi(2x) + _CI - \psi(2x)}{E \sqrt{\frac{\rho}{E}}}$$

Таким образом, решением задачи будут функции

> sigma:='sigma':sigma(x,t)=Sigma(x,t);

$$\sigma(x,t) = \frac{1}{2} \sqrt{\frac{\rho}{E}} E \phi\left(\frac{\sqrt{\frac{\rho}{E}} x + t}{\sqrt{\frac{\rho}{E}}}\right) + \frac{1}{2} \psi\left(\frac{\sqrt{\frac{\rho}{E}} x + t}{\sqrt{\frac{\rho}{E}}}\right) - \frac{1}{2} \sqrt{\frac{\rho}{E}} E \phi\left(x - \frac{t}{\sqrt{\frac{\rho}{E}}}\right) + \frac{1}{2} \psi\left(x - \frac{t}{\sqrt{\frac{\rho}{E}}}\right)$$

> w:='w':w(x,t)=collect(W(x,t),_C1);

$$w(x,t) = \frac{1}{2} \frac{-\sqrt{\frac{\rho}{E}} E \phi\left(\frac{\sqrt{\frac{\rho}{E}} x + t}{\sqrt{\frac{\rho}{E}}}\right) - \psi\left(\frac{\sqrt{\frac{\rho}{E}} x + t}{\sqrt{\frac{\rho}{E}}}\right)}{\sqrt{\frac{\rho}{E}} E} + \frac{1}{2} \frac{\sqrt{\frac{\rho}{E}} E \phi\left(x - \frac{t}{\sqrt{\frac{\rho}{E}}}\right) - \psi\left(x - \frac{t}{\sqrt{\frac{\rho}{E}}}\right)}{\sqrt{\frac{\rho}{E}} E}$$

то есть

$$\sigma(x,t) = \frac{\psi\left(x + \sqrt{\frac{E}{\rho}}t\right) + \psi\left(x - \sqrt{\frac{E}{\rho}}t\right)}{2} + \sqrt{\rho E} \frac{\phi\left(x + \sqrt{\frac{E}{\rho}}t\right) - \phi\left(x - \sqrt{\frac{E}{\rho}}t\right)}{2},$$

$$w(x, t) = \frac{\psi\left(x + \sqrt{\frac{E}{\rho}}t\right) - \psi\left(x - \sqrt{\frac{E}{\rho}}t\right)}{2\sqrt{\rho E}} + \frac{\varphi\left(x + \sqrt{\frac{E}{\rho}}t\right) + \varphi\left(x - \sqrt{\frac{E}{\rho}}t\right)}{2}.$$

Упражнения

Привести к каноническому виду следующие уравнения:

$$1. \frac{\partial^2 u}{\partial x^2} - 2 \frac{\partial^2 u}{\partial x \partial y} - 3 \frac{\partial^2 u}{\partial y^2} + \frac{\partial u}{\partial y} = 0.$$

Ответ. $\frac{\partial^2 u}{\partial \eta^2} - \frac{\partial^2 u}{\partial \xi^2} + \frac{1}{4} \frac{\partial u}{\partial \xi} = 0.$

$$2. \frac{\partial^2 u}{\partial x^2} - x \frac{\partial^2 u}{\partial y^2} = 0.$$

Ответ. $\frac{\partial^2 u}{\partial \eta^2} - \frac{\partial^2 u}{\partial \xi^2} + \frac{1}{3\eta} \frac{\partial u}{\partial \eta} = 0.$

$$3. \frac{\partial^2 u}{\partial x^2} - y \frac{\partial^2 u}{\partial y^2} = 0.$$

Ответ. $\frac{\partial^2 u}{\partial \eta^2} - \frac{\partial^2 u}{\partial \xi^2} + \frac{1}{\xi} \frac{\partial u}{\partial \xi} = 0.$

$$4. x \frac{\partial^2 u}{\partial x^2} - y \frac{\partial^2 u}{\partial y^2} = 0.$$

Ответ. $\frac{\partial^2 u}{\partial \eta^2} - \frac{\partial^2 u}{\partial \xi^2} + \frac{1}{\xi \eta} \left(\eta \frac{\partial u}{\partial \xi} - \xi \frac{\partial u}{\partial \eta} \right) = 0.$

$$5. x^2 \frac{\partial^2 u}{\partial x^2} + y^2 \frac{\partial^2 u}{\partial y^2} = 0.$$

Ответ. $\frac{\partial^2 u}{\partial \eta^2} + \frac{\partial^2 u}{\partial \xi^2} - \frac{\partial u}{\partial \xi} - \frac{\partial u}{\partial \eta} = 0.$

$$6. y^2 \frac{\partial^2 u}{\partial x^2} + x^2 \frac{\partial^2 u}{\partial y^2} = 0.$$

Ответ. $\frac{\partial^2 u}{\partial \eta^2} + \frac{\partial^2 u}{\partial \xi^2} + \frac{1}{2\xi\eta} \left(\eta \frac{\partial u}{\partial \xi} + \xi \frac{\partial u}{\partial \eta} \right) = 0.$

$$7. \quad y^2 \frac{\partial^2 u}{\partial x^2} - x^2 \frac{\partial^2 u}{\partial y^2} = 0.$$

$$\text{Ответ.} \quad \frac{\partial^2 u}{\partial \eta^2} - \frac{\partial^2 u}{\partial \xi^2} - \frac{1}{2\xi\eta} \left(\eta \frac{\partial u}{\partial \xi} - \xi \frac{\partial u}{\partial \eta} \right) = 0.$$

$$8. \quad (1+x^2) \frac{\partial^2 u}{\partial x^2} + (1+y^2) \frac{\partial^2 u}{\partial y^2} + y \frac{\partial u}{\partial y} = 0.$$

$$\text{Ответ.} \quad \frac{\partial^2 u}{\partial \eta^2} + \frac{\partial^2 u}{\partial \xi^2} - \text{th } \xi \frac{\partial u}{\partial \xi} = 0.$$

$$9. \quad \frac{\partial^2 u}{\partial x^2} - 2 \sin x \frac{\partial^2 u}{\partial x \partial y} + (2 - \cos^2 x) \frac{\partial^2 u}{\partial y^2} = 0.$$

$$\text{Ответ.} \quad \frac{\partial^2 u}{\partial \eta^2} + \frac{\partial^2 u}{\partial \xi^2} + \frac{\partial u}{\partial \eta} \cos \xi = 0.$$

$$10. \quad y^2 \frac{\partial^2 u}{\partial x^2} + 2y \frac{\partial^2 u}{\partial x \partial y} + \frac{\partial^2 u}{\partial y^2} = 0.$$

$$\text{Ответ.} \quad \frac{\partial^2 u}{\partial \eta^2} + \frac{2}{\xi + 2\eta} \frac{\partial u}{\partial \xi} = 0.$$

$$11. \quad x \frac{\partial^2 u}{\partial x^2} + 2x \frac{\partial^2 u}{\partial x \partial y} + (x-1) \frac{\partial^2 u}{\partial y^2} = 0.$$

$$\text{Ответ.} \quad \frac{\partial^2 u}{\partial \eta^2} - \frac{\partial^2 u}{\partial \xi^2} - \frac{1}{\eta} \frac{\partial u}{\partial \eta} = 0.$$

$$12. \quad \frac{\partial^2 u}{\partial x^2} - 2 \sin x \frac{\partial^2 u}{\partial x \partial y} - \cos^2 x \frac{\partial^2 u}{\partial y^2} - \cos x \frac{\partial u}{\partial y} = 0.$$

$$\text{Ответ.} \quad \frac{\partial^2 u}{\partial \eta^2} - \frac{\partial^2 u}{\partial \xi^2} = 0.$$

$$13. \quad (1+x^2)^2 \frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} + 2x(1+x^2) \frac{\partial u}{\partial x} = 0.$$

$$\text{Ответ.} \quad \frac{\partial^2 u}{\partial \eta^2} + \frac{\partial^2 u}{\partial \xi^2} = 0.$$

$$14. \quad \frac{\partial^2 u}{\partial x^2} - 2 \cos x \frac{\partial^2 u}{\partial x \partial y} - (3 + \sin^2 x) \frac{\partial^2 u}{\partial y^2} - y \frac{\partial u}{\partial y} = 0.$$

Ответ. $\frac{\partial^2 u}{\partial \xi \partial \eta} + \frac{\eta - \xi}{32} \left(\frac{\partial u}{\partial \xi} - \frac{\partial u}{\partial \eta} \right) = 0.$

15. $\frac{\partial^2 u}{\partial x^2} - 2x \frac{\partial^2 u}{\partial x \partial y} + x^2 \frac{\partial^2 u}{\partial y^2} - 2 \frac{\partial u}{\partial y} = 0.$

Ответ. $\frac{\partial^2 u}{\partial \eta^2} - \frac{\partial^2 u}{\partial \xi^2} = 0.$

16. $(1 + x^2) \frac{\partial^2 u}{\partial x^2} + (1 + y^2) \frac{\partial^2 u}{\partial y^2} + x \frac{\partial u}{\partial x} + y \frac{\partial u}{\partial y} = 0.$

Ответ. $\frac{\partial^2 u}{\partial \eta^2} + \frac{\partial^2 u}{\partial \xi^2} = 0.$

17. $\sin^2 x \frac{\partial^2 u}{\partial x^2} - 2y \sin x \frac{\partial^2 u}{\partial x \partial y} + y^2 \frac{\partial^2 u}{\partial y^2} = 0.$

Ответ. $\frac{\partial^2 u}{\partial \eta^2} + \frac{\xi}{\cos \eta - 1} \frac{\partial u}{\partial \xi} = 0.$

Следующие задачи решить с помощью формулы Даламбера:

18. Найти решение уравнения $\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}$, $-\infty < x < +\infty$, $t > 0$ удовлетворяющее начальным условиям $u|_{t=0} = \sin(x)$, $\frac{\partial u}{\partial t}|_{t=0} = 0.$

Ответ. $u = \sin x \cos t.$

19. Найти решение уравнения $\frac{\partial^2 u}{\partial t^2} = 4 \frac{\partial^2 u}{\partial x^2}$, $-\infty < x < +\infty$, $t > 0$, удовлетворяющее начальным условиям $u|_{t=0} = 0$, $\frac{\partial u}{\partial t}|_{t=0} = A \sin(x).$

Ответ. $u = A \sin x \sin t \cos t.$

20. Найти решение уравнения $\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}$, $-\infty < x < +\infty$, $t > 0$, если $u|_{t=0} = x^2$, $\frac{\partial u}{\partial t}|_{t=0} = 0.$

Ответ. $u = x^2 + t^2.$

21. Найти решение уравнения $\frac{\partial^2 u}{\partial t^2} = 4 \frac{\partial^2 u}{\partial x^2}$, $-\infty < x < +\infty$, $t > 0$, если $u|_{t=0} = 0$,

$$\left. \frac{\partial u}{\partial t} \right|_{t=0} = x.$$

Ответ. $u = xt$.

22. Найти форму струны, определяемой уравнением $\frac{\partial^2 u}{\partial t^2} = v^2 \frac{\partial^2 u}{\partial x^2}$, $-\infty < x < +\infty$,

$$t > 0 \text{ в момент времени } t = \pi/2v, \text{ если } u|_{t=0} = \sin(x), \left. \frac{\partial u}{\partial t} \right|_{t=0} = 1.$$

Ответ. $u = \sin x \cos vt + t$. Если $t = \pi/2v$, то $u = \pi/2v$, т.е. струна параллельна оси абсцисс.

23. Найти решение уравнения $\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}$, $-\infty < x < +\infty$, $t > 0$, если $u|_{t=0} = x$,

$$\left. \frac{\partial u}{\partial t} \right|_{t=0} = -x.$$

Ответ. $u = x - xt$.

24. Найти решение уравнения $\frac{\partial^2 u}{\partial t^2} = v^2 \frac{\partial^2 u}{\partial x^2}$, $-\infty < x < +\infty$, $t > 0$, если $u|_{t=0} = 0$,

$$\left. \frac{\partial u}{\partial t} \right|_{t=0} = \cos(x).$$

Ответ. $u = \frac{\cos x \sin vt}{v}$.

25. Найти форму струны, определяемой уравнением $\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}$, $-\infty < x < +\infty$,

$$t > 0 \text{ в момент времени } t = \pi, \text{ если } u|_{t=0} = \sin(x), \left. \frac{\partial u}{\partial t} \right|_{t=0} = \cos(x).$$

Ответ. $u = \sin(x + t)$. Если $t = \pi$, то $u = -\sin x$.

Найти общее решение уравнений с постоянными коэффициентами:

$$26. \frac{\partial^2 u}{\partial x \partial y} + a \frac{\partial u}{\partial x} + b \frac{\partial u}{\partial y} + abu = 0.$$

Ответ. $u = e^{-ay} f(x) + e^{-bx} g(y)$.

$$27. \frac{\partial^2 u}{\partial x^2} + 2a \frac{\partial^2 u}{\partial x \partial y} + a^2 \frac{\partial^2 u}{\partial y^2} + \frac{\partial u}{\partial x} + a \frac{\partial u}{\partial y} = 0.$$

Ответ. $u = f(y - ax) + e^{-x}g(y - ax)$.

$$28. \frac{\partial^2 u}{\partial x \partial y} + a \frac{\partial u}{\partial x} = 0.$$

Ответ. $u = f(y) + e^{-ay}g(x)$.

29. Доказать, что общее решение уравнения $\frac{\partial^2 u}{\partial x \partial y} - \frac{1}{x-y} \frac{\partial u}{\partial x} + \frac{1}{x-y} \frac{\partial u}{\partial y} = 0$ имеет

вид $u(x, y) = \frac{f(x) + g(y)}{x - y}$, где f и g произвольные функции из класса $C^{(2)}$.

30. Решить задачу Коши для системы

$$\begin{cases} 2 \frac{\partial u}{\partial t} - \frac{\partial u}{\partial x} - \frac{\partial v}{\partial x} = 0, \\ 2 \frac{\partial v}{\partial t} - \frac{\partial u}{\partial x} - \frac{\partial v}{\partial x} = 0, \end{cases}$$

$$u(x, 0) = 0, \quad v(x, 0) = 2x, \quad -\infty < x < \infty.$$

Ответ. $u = t, \quad v = 2x + t$.

4. Метод Фурье

В этой главе мы рассмотрим технику применения одного из основных аналитических методов решения краевых задач математической физики — метода Фурье. *Метод Фурье* или *метод разделения переменных* является одним из наиболее общих методов математической физики, применение которого не ограничивается уравнениями какого-либо одного типа (гиперболического, эллиптического или параболического) и оказывается эффективным при рассмотрении широкого класса задач.

Применение метода Фурье будет иметь успех, если уравнение задачи является линейным и относится к классу дифференциальных уравнений с разделяющимися переменными.

Уравнения с разделяющимися переменными

Как мы только что отметили, метод Фурье основан на разделении переменных и применяется для решения линейных дифференциальных уравнений в частных производных. Именно последние уравнения занимают особое место в общей теории дифференциальных уравнений с частными производными. Это объясняется тем, что, с одной стороны, они составляют наиболее разработанную часть этой теории, а с другой стороны, описывая реальные физические процессы, линейные дифференциальные уравнения в частных производных находят многочисленные применения в физике и технике. Решения этих уравнений обладают рядом замечательных свойств, напоминающих соответствующие свойства решений обыкновенных дифференциальных уравнений.

Напомним, что дифференциальное уравнение называется *линейным*, если неизвестная функция и все ее производные входят в это уравнение в первой степени и если оно не содержит членов с произведениями этих величин. Так, например, дифференциальное уравнение в частных производных второго порядка, линейное и *однородное* относительно неизвестной функции и ее производных, имеет вид

$$L(u) \equiv \sum_{i=1}^n \sum_{j=1}^n a_{ij}(x) \frac{\partial^2 u}{\partial x_i \partial x_j} + \sum_{i=1}^n b_i(x) \frac{\partial u}{\partial x_i} + c(x)u = 0, \quad (4.1)$$

где L — линейный дифференциальный оператор, определяемый выражением, стоящим в правой части уравнения; $u = u(x_1, x_2, \dots, x_n)$ — искомая функция; коэффициенты этого уравнения — заданные функции от $x = (x_1, x_2, \dots, x_n)$, где x — точка n -мерного евклидова пространства.

Если правая часть уравнения отлична от нуля, то есть, если мы имеем

$$L(u) = F(x),$$

где $F(x)$ — известная функция от $x = (x_1, x_2, \dots, x_n)$, то данное линейное уравнение называется *неоднородным*.

В основе важнейших методов интегрирования линейных уравнений в частных производных лежит общая для всех таких уравнений идея. Она заключается в

свойстве решений, которое называется *принципом суперпозиции (наложения)*. Этот принцип можно высказать в виде следующей теоремы.

Принцип суперпозиции. *Если каждая из функций*

$$u_1(x), u_2(x), \dots, u_n(x)$$

является решением однородного линейного дифференциального уравнения $L(u) = 0$, то и их линейная комбинация

$$u(x) = C_1 u_1(x) + C_2 u_2(x) + \dots + C_n u_n(x),$$

где C_1, C_2, \dots, C_n — произвольные постоянные, также является решением этого уравнения.

Действительно, достаточно проверить, что $L(u) = 0$, если

$$L(u_1) = 0, L(u_2) = 0, \dots, L(u_n) = 0.$$

Используя правила дифференцирования, легко получаем цепочку равенств

$$L(u) = C_1 L(u_1) + C_2 L(u_2) + \dots + C_n L(u_n) = 0,$$

что и требовалось доказать.

Принцип суперпозиции допускает следующее важное обобщение на случай бесконечных рядов.

Обобщенный принцип суперпозиции. *Если каждая из функций*

$$u_1(x), u_2(x), \dots, u_n(x), \dots$$

является решением однородного линейного дифференциального уравнения $L(u) = 0$, то ряд

$$u(x) = \sum_{n=1}^{\infty} C_n u_n(x)$$

также является решением этого уравнения, если он сходится к некоторой функции $u(x)$ и допускает почленное дифференцирование.

В самом деле, если все производные функции u , фигурирующие в уравнении $L(u) = 0$, вычисляются почленным дифференцированием ряда, то в силу линейности уравнения

$$L(u) = L\left(\sum_{n=1}^{\infty} C_n u_n\right) = \sum_{n=1}^{\infty} C_n L(u_n) = 0,$$

так как сходящиеся ряды можно складывать почленно. Тем самым доказано, что функция u удовлетворяет уравнению. В качестве достаточного условия для

возможности почленного дифференцирования ряда мы постоянно будем пользоваться условием равномерной сходимости ряда

$$\sum_{n=1}^{\infty} C_n L(u_n),$$

получаемого после дифференцирования.

Как известно, принцип суперпозиции, аналогичный сформулированному выше принципу для уравнения (4.1), имеет место и для обыкновенных линейных дифференциальных уравнений. Из теории этих уравнений известно также, что линейное уравнение n -го порядка имеет не более n линейно независимых решений. Поэтому в этом случае имеет смысл говорить лишь о линейной комбинации этих решений в числе не более n .

В случае же линейного уравнения в частных производных можно выбрать бесконечное множество линейно независимых частных решений¹. Это обстоятельство совместно с обобщенным принципом суперпозиции играет чрезвычайно важную роль в образовании решения уравнения в частных производных. Именно большой набор частных решений такого уравнения даст возможность выделить из общего решения то, которое должно удовлетворять, вообще говоря, произвольным дополнительным условиям.

Обобщение принципа суперпозиции идет еще дальше. Может оказаться, что решение уравнения в частных производных $L(u) = 0$ имеет вид $u(x, \xi)$, то есть зависит от некоторого параметра, изменяющегося в конечном или бесконечном промежутке (a, b) . Эта зависимость понимается так, что при всех значениях $\xi \in (a, b)$ функция $u(x, \xi)$ является решением уравнения, то есть $L(u(x, \xi)) \equiv 0$. Из предыдущего следует, что в свою очередь функция $v(\xi)u(x, \xi)$, где $v(\xi)$ — произвольная функция, будет также решением уравнения $L(u) = 0$. Как и ранее, интерес представляет бесконечная сумма выписанных решений. Но так как $v(\xi)$ зависит от непрерывно меняющегося параметра ξ , то это суммирование должно осуществляться при помощи интеграла. Таким образом, мы приходим к рассмотрению интеграла

$$\int_a^b v(\xi) u(x, \xi) d\xi.$$

Нетрудно убедиться, что этот интеграл является решением уравнения в частных производных $L(u) = 0$, конечно при условиях его существования и законности выполнения операции дифференцирования по x под знаком интеграла.

¹ Бесконечное множество функций называется линейно независимым, если любое конечное множество этих функций линейно независимо.

Рассмотрим однородное линейное уравнение (4.1): $L(u) = 0$. Мы скажем, что это уравнение относится к классу уравнений с разделяющимися переменными, если оно допускает бесконечное множество решений в виде

$$u = \prod_{i=1}^n X_i(x_i), \quad (4.2)$$

причем все решения $X_i(x_i)$ получаются из обыкновенных дифференциальных уравнений подстановкой (4.2) в (4.1).

Для того чтобы переменные в уравнении (4.1) разделялись необходимо, чтобы оператор L имел определенную структуру. В частности, переменные разделяются, если уравнение имеет вид

$$L(u) = L_1(u) + L_2(u) + \dots + L_n(u) = 0,$$

где L_1 — зависит только от x_1 , L_2 — зависит только от x_2 , L_n — зависит только от x_n . Например, если $n = 2$, то будем иметь

$$L(u) = L_1(u) + L_2(u) = 0, \quad (4.3)$$

Переменные разделяются, если оператор L имеет такой вид (в случае $n = 3$)

$$L(u) = L_1(u) + L_2(u) + g(x_2)L_3(u) = 0. \quad (4.4)$$

Рассмотрим подробнее методику разделения переменных на конкретных примерах. Если дано уравнение (4.1) с разделяющимися переменными, то все решения $X_i(x_i)$ в формуле (4.2) мы получим из обыкновенных дифференциальных уравнений путем подстановки (4.2) в (4.1). Пусть, например, уравнение имеет вид (4.3). Тогда это уравнение должно иметь решения вида $u(x_1, x_2) = X_1(x_1)X_2(x_2)$. Подставим это представление решения в уравнение (4.3), получим

$$X_2L_1(X_1) + X_1L_2(X_2) = 0 \Rightarrow \frac{L_1(X_1)}{X_1} + \frac{L_2(X_2)}{X_2} = 0, \quad (4.5)$$

В последнем уравнении одно слагаемое зависит только от x_1 , а другое — только от x_2 . Равенство (4.5) возможно лишь в случае, когда

$$\frac{L_1(X_1)}{X_1} = -\lambda, \quad \frac{L_2(X_2)}{X_2} = \lambda,$$

где $\lambda = const$. Таким образом, имеем обыкновенные дифференциальные уравнения

$$L_1(X_1) + \lambda X_1 = 0, \quad L_2(X_2) - \lambda X_2 = 0, \quad (4.6)$$

и разделение переменных осуществлено (число λ называется *параметром разделения*). Интегралы уравнений (4.6) зависят от параметра λ и постоянных интегрирования, то есть, получаем бесконечную совокупность частных решений.

Рассмотрим уравнение (4.4). Покажем, что это уравнение допускает решение в виде $u(x_1, x_2, x_3) = X_1(x_1)X_2(x_2)X_3(x_3)$. После подстановки этого представления в (4.4), получим

$$X_2X_3L_1(X_1) + X_1X_3L_2(X_2) + g(x_2)X_1X_2L_3(X_3) = 0,$$

или

$$\frac{L_1(X_1)}{X_1} + \frac{L_2(X_2)}{X_2} + \frac{g(x_2)L_3(X_3)}{X_3} = 0. \quad (4.7)$$

Равенство (4.7) возможно лишь в случае

$$\frac{L_1(X_1)}{X_1} = -\lambda, \quad \frac{L_2(X_2)}{X_2} + \frac{g(x_2)L_3(X_3)}{X_3} = \lambda.$$

Таким образом, получаем одно обыкновенное дифференциальное уравнение

$$L_1(X_1) + \lambda X_1 = 0.$$

Имеем далее,

$$\frac{L_2(X_2)}{g(x_2)X_2} - \frac{\lambda}{g(x_2)} + \frac{L_3(X_3)}{X_3} = 0. \quad (4.8)$$

Равенство (4.8) возможно лишь в случае, когда

$$\frac{L_3(X_3)}{X_3} = -\mu, \quad \frac{L_2(X_2)}{g(x_2)X_2} - \frac{\lambda}{g(x_2)} = \mu.$$

Откуда получаем еще два обыкновенных дифференциальных уравнения

$$L_3(X_3) + \mu X_3 = 0, \quad L_2(X_2) - [\mu g(x_2) + \lambda] X_2 = 0.$$

Иногда, чтобы добиться разделения переменных в некотором заданном уравнении, можно положить $u = f(x_1, x_2, \dots, x_n)w$. После подстановки такого выражения в уравнение $L(u) = 0$ с не разделяющимися переменными можно получить новое уравнение вида $M(w) = 0$, у которого переменные разделяются.

В задачах математической физики часто приходится сталкиваться с оператором Лапласа. Помимо уравнения Лапласа оператор Лапласа входит в уравнение теплопроводности, в волновое уравнение и в ряд других. Поэтому в случае, если в некоторой системе криволинейных координат переменные могут быть разделены в уравнении Лапласа, есть основания надеяться на то, что и многие более общие краевые задачи (в которые входит оператор Лапласа) могут быть решены методом Фурье. Самой простой и наиболее распространенной системой координат, в которой переменные в операторе Лапласа разделяются, является декартова прямоугольная система координат. В цилиндрической и сферической системах координат переменные в уравнении Лапласа также разделяются (читателю мы рекомендуем

убедиться в этом самостоятельно). Существует одиннадцать координатных систем, в которых возможно разделение переменных в операторе Лапласа. Использование специальных координатных систем может оказаться удобным в том случае, если граничные условия заданы на поверхностях, являющихся координатными поверхностями в соответствующей системе ортогональных координат. Так, например, если граничные условия заданы на сфере, удобно использовать сферические координаты, если границами являются цилиндрические и плоские области, следует перейти к цилиндрическим координатам, а если границей является сфероид, полезно использовать сфероидальную систему координат.

Метод Фурье для случая двух независимых переменных

Принцип метода Фурье состоит в том, что решение определенного класса задач математической физики, приводящихся к интегрированию уравнения $L(u) = 0$ при некоторых граничных и начальных условиях, строится при помощи суперпозиции частных решений рассматриваемого уравнения, имеющих вид произведения множителей, каждый из которых является функцией только одной переменной, и удовлетворяющих однородным условиям задачи.

При определенной структуре оператора L попытка искать решение уравнения $L(u) = 0$ в форме произведения указанного вида приводит к разделению переменных, и нахождение каждого множителя сводится к интегрированию обыкновенного дифференциального уравнения, что представляет относительно простую задачу.

Посредством надлежащего выбора значений части параметров, входящих в выражения найденных таким образом интегралов, обычно оказывается возможным удовлетворить всем однородным условиям, что приводит к выделению счетной или несчетной совокупности частных решений требуемого вида. Решение задачи получается путем суперпозиции построенных решений, причем выполнение неоднородных условий, наложенных на искомую функцию, осуществляется с помощью соответствующего выбора оставшейся части параметров.

Сформулируем следующую задачу. Пусть требуется найти функцию $u = u(x, y)$, которая удовлетворяет следующим условиям: линейному дифференциальному уравнению

$$L_x(u) + M_y(u) = 0, \quad a < x < b, \quad c < y < d. \quad (4.9)$$

где

$$L_x(u) = \frac{1}{r(x)} \left[\frac{\partial}{\partial x} \left(p(x) \frac{\partial u}{\partial x} \right) - q(x)u \right]$$

$$M_y(u) = A \frac{\partial^2 u}{\partial y^2} + B \frac{\partial u}{\partial y} + Cu$$

Функции p, q, r — вещественные функции от x ; A, B, C — вещественные функции от y . Предполагается, что p, p', q, r непрерывны на $[a, b]$; $p > 0, r > 0$; A, B, C

непрерывны на (c, d) (интервал (c, d) может быть и бесконечным). Мы сначала предполагаем, что интервал $[a, b]$ конечен и поведение рассматриваемых на концах этого интервала функций таково, что каждое из отношений

$$\frac{p'(x)}{p(x)}, \frac{q(x)}{p(x)}, \frac{r(x)}{p(x)} \quad (4.10)$$

стремится к конечному пределу, когда $x \rightarrow a$ и $x \rightarrow b$.

По переменной x выполняется одно из однородных граничных условий I, II или III рода, то есть

$$u|_{x=a} = 0, \quad u|_{x=b} = 0, \quad (4.11-I)$$

либо

$$\frac{\partial u}{\partial x}\bigg|_{x=a} = 0, \quad \frac{\partial u}{\partial x}\bigg|_{x=b} = 0, \quad (4.11-II)$$

либо

$$\frac{\partial u}{\partial x} - h_a u\bigg|_{x=a} = 0, \quad \frac{\partial u}{\partial x} + h_b u\bigg|_{x=b} = 0, \quad h_a > 0, \quad h_b > 0. \quad (4.11-III)$$

По переменной y выполняются определенные неоднородные условия, которые зависят от типа уравнения (4.9). Квадратичная форма, связанная с уравнением (4.9), имеет вид

$$G = \frac{p}{r} X^2 + AY^2.$$

Тип уравнения определяется знаком A . Если $A > 0$, то уравнение (4.9) эллиптического типа и условия на концах интервала (c, d) имеют характер условий I, II или III рода, например,

$$u|_{y=c} = f_c(x), \quad u|_{y=d} = f_d(x).$$

Если $A < 0$, то уравнение (4.9) гиперболического типа. Этот тип часто встречается в случае, когда переменная y — время, $y \in (c, +\infty)$. Условия имеют вид

$$u|_{y=c} = \varphi(x), \quad \frac{\partial u}{\partial y}\bigg|_{y=c} = \psi(x).$$

Если $A = 0$, то уравнение (4.9) параболического типа; переменная y — время, $y \in (c, +\infty)$. Условия имеют вид

$$u|_{y=c} = \varphi(x).$$

Таким образом, уравнение (4.9) должно быть однородным и допускать разделение переменных; условия по переменной x — однородные, по переменной y — не обязательно.

Будем искать решение задачи в виде

$$u = X(x)Y(y). \quad (4.12)$$

После подстановки (4.12) в (4.9) получим

$$YL_x(X) + XM_y(Y) = 0 \Rightarrow \frac{L_x(X)}{X} = -\frac{M_y(Y)}{Y} = -\lambda.$$

Таким образом, уравнение (4.9) распадается на два обыкновенных дифференциальных уравнения

$$L_x(X) + \lambda X = 0, \quad (4.13)$$

$$M_y(Y) - \lambda Y = 0 \quad (4.14)$$

Уравнение (4.13) в явном виде записывается так

$$(pX')' + (\lambda r - q)X = 0. \quad (4.15)$$

Это уравнение Штурма–Лиувилля. В соответствии с граничными условиями (4.11) для функции $X(x)$ должно выполняться одно из трех условий

$$X|_{x=a} = 0, \quad X|_{x=b} = 0, \quad (4.16-I)$$

либо

$$\frac{dX}{dx}\Big|_{x=a} = 0, \quad \frac{dX}{dx}\Big|_{x=b} = 0, \quad (4.16-II)$$

либо

$$\frac{dX}{dx} - h_a X\Big|_{x=a} = 0, \quad \frac{dX}{dx} + h_b X\Big|_{x=b} = 0, \quad h_a > 0, \quad h_b > 0. \quad (4.16-III)$$

Таким образом, мы имеем дело с *регулярной задачей Штурма–Лиувилля* (4.15), (4.16). Такую задачу мы уже рассматривали в главе 2. Напомним здесь некоторые факты. При произвольных λ эта задача не имеет решения, за исключением тривиального $X \equiv 0$. Значение λ , при котором существует нетривиальное решение рассматриваемой задачи, называется *собственным значением* или *собственным числом*, а соответствующее решение — *собственной функцией*. Совокупность собственных значений образует *спектр* данной задачи.

Согласно общей теории (см., например, [3]), задача имеет *дискретный спектр*, состоящий из счетного множества вещественных собственных значений $\lambda = \lambda_n$, причем $n = 1, 2, 3, \dots, \infty$. Каждому собственному значению соответствует собственная

функция $X = X_n(x)$, определенная с точностью до произвольного постоянного множителя. Эти собственные функции ортогональны на $[a, b]$ с весом $r(x)$, то есть

$$\int_a^b r(x) X_m(x) X_n(x) dx = \begin{cases} 0, & m \neq n; \\ \|X_n\|^2, & m = n. \end{cases}$$

Собственные функции могут быть использованы для представления произвольных функций, определенных в данном промежутке. При наличии некоторых ограничений² функция $f(x)$, заданная на (a, b) , может быть разложена в ряд по собственным функциям

$$f(x) = \sum_{n=1}^{\infty} f_n X_n(x), \quad a < x < b$$

с коэффициентами

$$f_n = \frac{\int_a^b r(x) f(x) X_n(x) dx}{\int_a^b r(x) X_n^2(x) dx}.$$

Фактическое вычисление собственных значений и собственных функций легко выполняется в случае, если при произвольном λ известна система линейно независимых решений уравнения (4.15). Подстановка выражения общего интеграла этого уравнения в граничные условия (4.16) приводит в этом случае к однородной линейной системе для постоянных, и из условия равенства нулю определителя этой системы непосредственно получается трансцендентное уравнение для отыскания допустимых значений λ . После определения собственных значений и собственных функций второй множитель $Y(y)$ находится из уравнения (4.14) (с $\lambda = \lambda_n$).

Подставляя собственные значения $\lambda = \lambda_n$ в уравнение (4.14) получим

$$M_y(Y) - \lambda_n Y = 0. \quad (4.17)$$

Для определенности будем считать, что $A \neq 0$. Тогда общий интеграл уравнения (4.17) имеет вид³

$$Y_n(y) = C_n Y_n^{(1)}(y) + D_n Y_n^{(2)}(y),$$

² Например, при условии, что $f(x)$ — кусочно-гладкая функция в интервале (a, b)

³ При $A = 0$ соответствующая форма решения будет: $Y(y) = C_n Y_n(y)$.

где $Y_n^{(1)}(y)$, $Y_n^{(2)}(y)$ — линейно независимые решения уравнения (4.17), то есть их определитель Вронского отличен от нуля

$$W\{Y_n^{(1)}, Y_n^{(2)}\} \neq 0.$$

Таким образом, мы приходим к следующей бесконечной совокупности частных решений уравнения (4.9)

$$u = u_n = [M_n Y_n^{(1)}(y) + N_n Y_n^{(2)}(y)] X_n(x), \quad n = 1, 2, 3, \dots, \infty.$$

Каждое из этих решений удовлетворяет уравнению (4.9) и однородным граничным условиям по переменной x . Для того чтобы удовлетворить неоднородным условиям по переменной y , мы применяем принцип суперпозиции и составляем решение в виде ряда

$$u = \sum_{n=1}^{\infty} u_n = \sum_{n=1}^{\infty} [M_n Y_n^{(1)}(y) + N_n Y_n^{(2)}(y)] X_n(x). \quad (4.18)$$

Мы постулируем здесь сходимость ряда (4.18), допускающую предельный переход и дифференцирование под знаком суммы. Таким образом, можно написать

$$u|_{y=c} = \varphi(x) = \sum_{n=1}^{\infty} [M_n Y_n^{(1)}(c) + N_n Y_n^{(2)}(c)] X_n(x), \quad a < x < b, \quad (4.19)$$

$$\left. \frac{\partial u}{\partial y} \right|_{y=c} = \psi(x) = \sum_{n=1}^{\infty} [M_n Y_n^{(1)'}(c) + N_n Y_n^{(2)'}(c)] X_n(x), \quad a < x < b. \quad (4.20)$$

Равенства (4.19) и (4.20) означают разложения функций φ и ψ по собственным функциям задачи Штурма–Лиувилля (4.15)–(4.16). Если имеет место такое разложение, то можно написать

$$M_n Y_n^{(1)}(c) + N_n Y_n^{(2)}(c) = \frac{\int_a^b r(x) \varphi(x) X_n(x) dx}{\|X_n\|^2} = \varphi_n, \quad (4.21)$$

$$M_n Y_n^{(1)'}(c) + N_n Y_n^{(2)'}(c) = \frac{\int_a^b r(x) \psi(x) X_n(x) dx}{\|X_n\|^2} = \psi_n. \quad (4.22)$$

Таким образом, мы имеем систему уравнений (4.21) и (4.22) для определения неизвестных констант M_n , N_n . Определитель этой системы есть определитель Вронского, который отличен от нуля в силу линейной независимости функций $Y_n^{(1)}(y)$, $Y_n^{(2)}(y)$

$$\Delta = \begin{vmatrix} Y_n^{(1)}(c) & Y_n^{(2)}(c) \\ Y_n^{(1)'}(c) & Y_n^{(2)'}(c) \end{vmatrix} = W\{Y_n^{(1)}, Y_n^{(2)}\} \neq 0.$$

Следовательно, система разрешима относительно M_n, N_n и формула (4.18) дает формальное решение задачи.

Укажем краткую схему решения задачи методом Фурье:

1. Произвести разделение переменных и свести задачу к интегрированию двух обыкновенных дифференциальных уравнений.
2. Сформулировать задачу Штурма–Лиувилля и найти совокупность собственных значений и собственных функций рассматриваемой задачи.
3. Проинтегрировать второе уравнение и построить совокупность частных решений задачи, удовлетворяющих однородным граничным условиям.
4. Составить формальный ряд из частных решений и, пользуясь теоремой разложения по собственным функциям, найти коэффициенты разложения.
5. Дать строгое обоснование полученному формальному решению для определенного класса функций.

Данная схема остается в силе для случая условий IV рода (условий периодичности) регулярной задачи Штурма–Лиувилля:

$$u|_{x=a} = u|_{x=b}, \quad \frac{\partial u}{\partial x}|_{x=a} = \frac{\partial u}{\partial x}|_{x=b}.$$

Единственное отличие состоит в том, что в этом случае одному собственному значению $\lambda = \lambda_n$ могут соответствовать две линейно независимые собственные функции.

Некоторые трудности представляет распространение метода Фурье на случаи, когда интервал (a, b) бесконечен или какое-либо из отношений (4.10) обращается в бесконечность при приближении к одному или обоим концам рассматриваемого интервала (случай одного или двух сингулярных концов). В этих случаях граничное условие, относящееся к сингулярному концу или к концу $x = b = \infty$, вообще говоря, не может быть задано в произвольной форме и заменяется граничным условием, формулировка которого в конкретных задачах математической физики обычно не представляет особых затруднений. Чаще всего это условие заключается в требовании ограниченности решения при приближении к сингулярному концу.

Собственные функции в задачах с конечным интервалом изменения переменной x и с сингулярностью на одном конце определяются как нетривиальные решения уравнения (4.15), удовлетворяющие на сингулярном конце упомянутому условию и условию типа (4.16) на другом конце интервала.

Аналогичным образом определяются собственные функции в случае конечного интервала с двумя сингулярными концами, для бесконечного интервала и т. д.

Существенное отличие рассматриваемых случаев от разобранных выше состоит в том, что здесь спектр собственных значений может быть как дискретным, так и недискретным, в зависимости от структуры уравнения и характера налагаемых условий⁴.

Приведенная выше схема решения задачи методом Фурье также переносится и на сингулярный случай, если спектр задачи оказывается дискретным. Если спектр задачи будет непрерывным или смешанным, то возникают некоторые изменения в схеме. В этом случае меняется сам характер решения, которое строится из частных решений путем интегрирования по параметру λ , причем для определения неизвестных множителей, входящих под знак интеграла, служит аппарат интегральных разложений, заменяющий теорию рядов по собственным функциям.

Примеры решения типовых задач

Задача 4.1. Решить задачу для уравнения теплопроводности

$$\frac{\partial u}{\partial t} - a^2 \frac{\partial^2 u}{\partial x^2} = 0, \quad 0 < x < L, \quad t > 0$$

с начальным

$$u(x, 0) = \frac{cx(L-x)}{L}$$

и граничными условиями

$$u(0, t) = 0, \quad u(L, t) = 0.$$

Решение. Эта задача может быть решена методом Фурье. Мы воспользуемся для ее решения системой Maple.

Определяем задачу в Maple:

```
> eq:=diff(u(x,t),t)-a^2*diff(u(x,t),x,x)=0;
```

$$eq := \left(\frac{\partial}{\partial t} u(x, t) \right) - a^2 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) = 0$$

```
> init_c:=u(x,0)=phi(x);
```

$$init_c := u(x, 0) = \phi(x)$$

```
> bound_c:=u(0,t)=0,u(L,t)=0;
```

$$bound_c := u(0, t) = 0, u(L, t) = 0$$

```
> phi:=x->c*x*(L-x)/L^2;
```

$$\phi := x \rightarrow \frac{c x (L - x)}{L^2}$$

⁴ Примерами задач с недискретным спектром могут служить задачи с непрерывным спектром, рассмотренные в главе 6.

Выполняем разделение переменных:

```
> res:=pdsolve(eq,HINT=X(x)*T(t));
      res := (u(x, t) = X(x) T(t)) &where
      [ {  $\frac{d}{dt} T(t) = a^2 \_c_1 T(t), \frac{d^2}{dx^2} X(x) = \_c_1 X(x) \} ]$ 
```

```
> res1:=op(2,res);
      res1 := [ {  $\frac{d}{dt} T(t) = a^2 \_c_1 T(t), \frac{d^2}{dx^2} X(x) = \_c_1 X(x) \} ]$ 
```

```
> s1:=op(1,res1[1]);s2:=op(2,res1[1]);
      s1 :=  $\frac{d^2}{dx^2} X(x) = \_c_1 X(x)$ 
      s2 :=  $\frac{d}{dt} T(t) = a^2 \_c_1 T(t)$ 
```

Таким образом, мы получили два обыкновенных дифференциальных уравнения. Сформулируем теперь соответствующую задачу Штурма–Лиувилля. Мы имеем однородные условия по переменной x . Поэтому рассматриваем уравнение $s1$, которое удобно записать в виде

```
> eq1:=lhs(s1)+lambda*X(x);
      eq1 :=  $\left( \frac{d^2}{dx^2} X(x) \right) + \lambda X(x)$ 
```

Находим общее решение этого уравнения и формируем систему однородных уравнений по граничным условиям:

```
> dsolve(eq1,X(x));
      X(x) =  $\_C1 \sin(\sqrt{\lambda} x) + \_C2 \cos(\sqrt{\lambda} x)$ 
```

```
> X:=unapply(rhs(%),x);
      e1:=X(0)=0; e2:=X(L)=0;
      sist:={e1,e2};
      X:=x →  $\_C1 \sin(\sqrt{\lambda} x) + \_C2 \cos(\sqrt{\lambda} x)$ 
      e1 :=  $\_C2 = 0$ 
      e2 :=  $\_C1 \sin(\sqrt{\lambda} L) + \_C2 \cos(\sqrt{\lambda} L) = 0$ 
      sist := {  $\_C1 \sin(\sqrt{\lambda} L) + \_C2 \cos(\sqrt{\lambda} L) = 0, \_C2 = 0 \}$ 
```

Вычисляем определитель этой системы:

```
> A:=linalg[genmatrix](sist,{_C1,_C2});
      A :=  $\begin{bmatrix} \sin(\sqrt{\lambda} L) & \cos(\sqrt{\lambda} L) \\ 0 & 1 \end{bmatrix}$ 
```

```
> Delta:=convert(linalg[det](A),trig);
```

$$\Delta := \sin(\sqrt{\lambda} L)$$

Приравниваем определитель нулю — это будет уравнение для определения собственных значений; решаем полученное уравнение и находим собственные значения задачи:

```
> _EnvAllSolutions := true:
```

```
> solve(Delta,lambda);
```

$$\frac{\pi^2 _Z1^2}{L^2}$$

```
> subs(_Z1='k',%):
```

```
> ev:=unapply(%,k);
```

$$ev := k \rightarrow \frac{\pi^2 k^2}{L^2}$$

Определяем теперь соответствующие собственные функции:

```
> X:='X':assume(k,posint):subs(lambda=ev(k),eq1);
```

$$\left(\frac{d^2}{dx^2} X(x) \right) + \frac{\pi^2 k^2 X(x)}{L^2}$$

```
> dsolve({%,X(0)=0,X(L)=0},X(x));
```

$$X(x) = _C1 \sin\left(\frac{\pi k x}{L}\right)$$

Нормируем эти функции:

```
> rhs(%) / sqrt(int(rhs(%)^2, x=0..L));
```

$$\frac{_C1 \sin\left(\frac{\pi k x}{L}\right) \sqrt{2}}{\sqrt{_C1^2 L}}$$

```
> simplify(% , radical, symbolic):
```

```
> ef:=unapply(%,(k,x));
```

$$ef := (k, x) \rightarrow \frac{\sin\left(\frac{\pi k x}{L}\right) \sqrt{2}}{\sqrt{L}}$$

Итак, собственные значения и нормированные собственные функции задачи Штурма–Лиувилля найдены и выражаются по формулам:

```
> ev(k); ef(k,x);
```

$$\frac{\pi^2 k^2}{L^2}$$

$$\frac{\sin\left(\frac{\pi k x}{L}\right) \sqrt{2}}{\sqrt{L}}$$

Читателю мы рекомендуем самостоятельно убедиться в том, что $\lambda = 0$ не является собственным значением рассматриваемой задачи Штурма–Лиувилля.

Найдем теперь общее решение второго дифференциального уравнения s^2 :

> eq2 := lhs(s2) + a^2 * ev(k) * T(t) ;

$$eq2 := \left(\frac{d}{dt} T(t)\right) + \frac{a^2 \pi^2 k^2 T(t)}{L^2}$$

> dsolve(eq2, T(t)) ;

$$T(t) = _C1 e^{\left(-\frac{a^2 \pi^2 k^2 t}{L^2}\right)}$$

Решение исходной задачи ищем в виде ряда:

> U := (x, t) ->

Sum(C(k) * exp(-ev(k) * a^2 * t) * ef(k, x), k=1..infinity) ;

'u(x, t)' = U(x, t) ;

$$u(x, t) = \sum_{k=1}^{\infty} \frac{C(k) e^{\left(-\frac{a^2 \pi^2 k^2 t}{L^2}\right)} \sin\left(\frac{\pi k x}{L}\right) \sqrt{2}}{\sqrt{L}}$$

Коэффициенты этого ряда определяем согласно теореме разложения из начальных условий

> value(subs(u=U, init_c)) ;

$$\sum_{k=1}^{\infty} \frac{C(k) \sin\left(\frac{\pi k x}{L}\right) \sqrt{2}}{\sqrt{L}} = \frac{c x (L - x)}{L^2}$$

Для коэффициентов Фурье будем иметь:

> Ck := Int((phi(x)) * ef(k, x), x=0..L) ; Ck := value(Ck) ;

$$Ck := \int_0^L \frac{c x (L - x) \sin\left(\frac{\pi k x}{L}\right) \sqrt{2}}{L^{(5/2)}} dx$$

$$Ck := -\frac{2 \sqrt{L} c \sqrt{2} ((-1)^k - 1)}{\pi^3 k^3}$$

> C := unapply(Ck, k) ;

$$C := k \rightarrow -\frac{2\sqrt{L} c \sqrt{2} ((-1)^k - 1)}{\pi^3 k^3}$$

Итак, мы получили формальное решение задачи:

> 'u(x, t) '=U(x, t) ;

$$u(x, t) = \sum_{k=1}^{\infty} \left(-\frac{4 c ((-1)^k - 1) e^{\left(-\frac{a^2 \pi^2 k^2 t}{L^2}\right)} \sin\left(\frac{\pi k x}{L}\right)}{\pi^3 k^3} \right)$$

Преобразуем это решение к другому виду. Индекс k заменим на $2m+1$:

> subs(k=2*m+1, op(1, U(x, t))) ; assume(m, integer) :

$$-\frac{4 c ((-1)^{(2 m+1)} - 1) e^{\left(-\frac{a^2 \pi^2 (2 m+1)^2 t}{L^2}\right)} \sin\left(\frac{\pi (2 m+1) x}{L}\right)}{\pi^3 (2 m+1)^3}$$

> simplify(%):factor(%);

$$\frac{8 c e^{\left(-\frac{a^2 \pi^2 (2 m+1)^2 t}{L^2}\right)} \sin\left(\frac{\pi (2 m+1) x}{L}\right)}{\pi^3 (2 m+1)^3}$$

> 'u(x, t) '=Sum(%, m=0..infinity) ;

$$u(x, t) = \sum_{m=0}^{\infty} \left(\frac{8 c e^{\left(-\frac{a^2 \pi^2 (2 m+1)^2 t}{L^2}\right)} \sin\left(\frac{\pi (2 m+1) x}{L}\right)}{\pi^3 (2 m+1)^3} \right)$$

Таким образом, мы окончательно получили решение задачи в виде

$$u(x, t) = \frac{8c}{\pi^3} \sum_{m=0}^{\infty} \frac{e^{\left(-\frac{a^2 \pi^2 (2m+1)^2 t}{L^2}\right)} \sin\left(\frac{\pi (2m+1) x}{L}\right)}{(2m+1)^3}.$$

Задача 4.2. Рассмотреть продольные колебания стержня, конец которого $x=0$ жестко закреплен, а свободный конец $x=l$ получает в начальный момент времени продольный ударный импульс P . До удара стержень находился в состоянии покоя.

Решение. Поставим задачу — найти решение уравнения колебаний

$$\frac{\partial^2 u}{\partial x^2} - \frac{1}{v^2} \frac{\partial^2 u}{\partial t^2} = 0, \quad 0 < x < l, \quad t > 0,$$

подчиняющееся граничным условиям

$$u|_{x=0} = 0, \quad \left. \frac{\partial u}{\partial x} \right|_{x=l} = 0$$

и начальным условиям. Начальные условия мы можем записать в следующем виде

$$u|_{t=0} = 0, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = \begin{cases} 0, & 0 < x < l - \varepsilon; \\ v_0 = \frac{P}{\rho \varepsilon S}, & l - \varepsilon < x < l. \end{cases}$$

Здесь S — площадь поперечного сечения стержня, ρ — плотность.

После того, как решение сформулированной задачи будет получено, необходимо перейти к пределу при $\varepsilon \rightarrow 0$.

Можно начальные условия сформулировать в другом виде, использующем понятие дельта-функции Дирака:

$$u|_{t=0} = 0, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = v_0(x) = \frac{P}{\rho S} \delta(x-l).$$

Коэффициент при дельта-функции $\delta(x-l)$ выбирается так, чтобы суммарный импульс, передаваемый стержню в момент $t=0$, был равен P , то есть

$$\int_0^l \frac{\partial u(x,0)}{\partial t} \rho S dx = P.$$

Рассмотрим обе возможности задания начальных условий. Для решения задачи воспользуемся Maple. Поставим задачу — дифференциальное уравнение колебаний:

$$\text{> eq:=diff(u(x,t),t,t)/v^2-diff(u(x,t),x,x)=0;$$

$$\text{eq} := \frac{\partial^2 u(x,t)}{\partial t^2} - \left(\frac{\partial^2 u(x,t)}{\partial x^2} \right) = 0$$

Начальные условия:

$$\text{> init_c:=u(x,0)=0,D[2](u)(x,0)=phi(x);$$

$$\text{init_c} := u(x,0) = 0, D_2(u)(x,0) = \phi(x)$$

Функцию $\phi(x)$ определим двумя способами:

$$\text{> phi1:=piecewise}(0 < x \text{ and } x < l - \text{epsilon}, 0,$$

$$l - \text{epsilon} < x \text{ and } x < l, P / \rho / \text{epsilon} / S);$$

$$\phi 1 := \begin{cases} 0 & -x < 0 \text{ and } x - l + \varepsilon < 0 \\ \frac{P}{\rho \varepsilon S} & -x + l - \varepsilon < 0 \text{ and } x - l < 0 \end{cases}$$

или

> phi2:=x->P/rho/S*Dirac(x-l);

$$\phi_2 := x \rightarrow \frac{P \operatorname{Dirac}(x-l)}{\rho S}$$

Граничные условия:

> bound_c:=u(0,t)=0,D[1](u)(0,t)=0;

$$\text{bound_c} := u(0, t) = 0, D_1(u)(0, t) = 0$$

Выполним разделение переменных в уравнении колебаний:

> subs(u(x,t)=X(x)*T(t),eq);

$$\frac{\frac{\partial^2}{\partial t^2}(X(x) T(t))}{v^2} - \left(\frac{\partial^2}{\partial x^2}(X(x) T(t)) \right) = 0$$

> expand(lhs(%)/X(x)/T(t))=0;

$$\frac{\frac{d^2}{dt^2} T(t)}{T(t) v^2} - \frac{\frac{d^2}{dx^2} X(x)}{X(x)} = 0$$

> s1:=op(1,lhs(%))=-lambda;s2:=op(2,lhs(%))=lambda;

$$s1 := \frac{\frac{d^2}{dt^2} T(t)}{T(t) v^2} = -\lambda$$

$$s2 := -\frac{\frac{d^2}{dx^2} X(x)}{X(x)} = \lambda$$

Таким образом, мы получили два обыкновенных дифференциальных уравнения. Обратите внимание, здесь мы выполнили разделение переменных «вручную» без использования команды pdsolve.

Сформулируем теперь соответствующую задачу Штурма–Лиувилля. Мы имеем однородные условия по переменной x . Поэтому рассматриваем уравнение $s2$. Находим общее решение этого уравнения и формируем систему однородных уравнений по граничным условиям:

> assume(lambda>0):dsolve(s2,X(x));

$$X(x) = _C1 \sin(\sqrt{\lambda} x) + _C2 \cos(\sqrt{\lambda} x)$$

> X:=unapply(rhs(%),x);

$$X := x \rightarrow _C1 \sin(\sqrt{\lambda} x) + _C2 \cos(\sqrt{\lambda} x)$$

> e1:=X(0)=0; e2:=D(X)(l)=0;

sist:={e1,e2};

$$\begin{aligned}
e1 &:= _C2 = 0 \\
e2 &:= _C1 \cos(\sqrt{\lambda\sim} l) \sqrt{\lambda\sim} - _C2 \sin(\sqrt{\lambda\sim} l) \sqrt{\lambda\sim} = 0 \\
sist &:= \{ _C2 = 0, _C1 \cos(\sqrt{\lambda\sim} l) \sqrt{\lambda\sim} - _C2 \sin(\sqrt{\lambda\sim} l) \sqrt{\lambda\sim} = 0 \}
\end{aligned}$$

Вычисляем определитель этой системы:

```
> A:=linalg[genmatrix](sist, {_C1, _C2});
```

$$A := \begin{bmatrix} 0 & 1 \\ \cos(\sqrt{\lambda\sim} l) \sqrt{\lambda\sim} & -\sin(\sqrt{\lambda\sim} l) \sqrt{\lambda\sim} \end{bmatrix}$$

```
> Delta:=convert(linalg[det](A), trig);
```

$$\Delta := -\cos(\sqrt{\lambda\sim} l) \sqrt{\lambda\sim}$$

```
> Delta:=op(2, Delta);
```

$$\Delta := \cos(\sqrt{\lambda\sim} l)$$

Приравниваем определитель нулю — это будет уравнение для определения собственных значений; решаем полученное уравнение и находим собственные значения задачи:

```
> _EnvAllSolutions:= true:
```

```
solve(Delta, lambda);
```

$$\frac{\pi^2 (1 + 2 _Z1\sim)^2}{4 l^2}$$

```
> subs(_Z1='k', %):
```

```
ev:=unapply(%, k);
```

$$ev := k \rightarrow \frac{1}{4} \frac{\pi^2 (1 + 2 k)^2}{l^2}$$

Определяем теперь соответствующие собственные функции:

```
> assume(k, posint);
```

```
X:='X'; subs(lambda=ev(k), s2);
```

```
X:=X
```

$$-\frac{\frac{d^2}{dx^2} X(x)}{X(x)} = \frac{\pi^2 (1 + 2 k)^2}{4 l^2}$$

```
> dsolve({%, X(0)=0, D(X)(l)=0}, X(x));
```

$$X(x) = _C1 \sin\left(\frac{\pi (1 + 2 k) x}{2 l}\right)$$

Нормируем эти функции:

```
> rhs(%) / sqrt(int(rhs(%)^2, x=0..l));
```

$$\frac{-Cl \sin\left(\frac{\pi(1+2k^-)x}{2l}\right)\sqrt{2}}{\sqrt{-Cl^2l}}$$

> simplify(% , radical , symbolic) :

ef:=unapply(% , (k , x)) :

Итак, собственные значения и нормированные собственные функции задачи Штурма–Лиувилля найдены:

> ev(k) ; ef(k , x) ;

$$\frac{\pi^2(1+2k^-)^2}{4l^2}$$

$$\frac{\sin\left(\frac{\pi(1+2k^-)x}{2l}\right)\sqrt{2}}{\sqrt{l}}$$

Читателю мы рекомендуем самостоятельно убедиться в том, что $\lambda = 0$ не является собственным значением рассматриваемой задачи Штурма–Лиувилля.

Найдем теперь общее решение второго дифференциального уравнения s1:

> s1:=lhs(s1)=-ev(k) ;

$$s1 := \frac{d^2 T(t)}{dt^2 v^2} = -\frac{\pi^2(1+2k^-)^2}{4l^2}$$

> dsolve({s1 , T(0)=0} , T(t)) ;

$$T(t) = -Cl \sin\left(\frac{\pi v(1+2k^-)t}{2l}\right)$$

Здесь мы сразу учли одно из начальных условий. Решение исходной задачи будем строить в виде ряда:

> spr:=Sum(C(k)*op(2,rhs(%))*ef(k,x),k=1..infinity) ;

$$spr := \sum_{k=1}^{\infty} \frac{C(k^-) \sin\left(\frac{\pi v(1+2k^-)t}{2l}\right) \sin\left(\frac{\pi(1+2k^-)x}{2l}\right)\sqrt{2}}{\sqrt{l}}$$

Определим теперь неизвестные константы в формуле решения. Рассмотрим первый случай задания начальных условий — с предельным переходом. Пользуясь соответствующей теоремой разложения и учитывая второе начальное условие (на производную), будем иметь формулу для коэффициентов Фурье:

> assume(l>0) :

> Ck1:=P/rho/epsilon/S*

> Int(ef(k,x),x=l-epsilon..l)/ev(k)^(1/2)/v ;

$$Ck1 := \frac{P \sqrt{4}}{\rho \varepsilon S \sqrt{\frac{\pi^2 (1 + 2 k\sim)^2}{l\sim^2}} v} \int_{l\sim - \varepsilon}^{l\sim} \frac{\sin\left(\frac{\pi (1 + 2 k\sim) x}{2 l\sim}\right) \sqrt{2}}{\sqrt{l\sim}} dx$$

> Ck1 := simplify(value(Ck1));

$$Ck1 := \frac{4 P \sqrt{2} l\sim^{(3/2)} (-1)^{k\sim} \left(\sin\left(\frac{\pi \varepsilon}{2 l\sim}\right) \cos\left(\frac{\pi \varepsilon k\sim}{l\sim}\right) + \cos\left(\frac{\pi \varepsilon}{2 l\sim}\right) \sin\left(\frac{\pi \varepsilon k\sim}{l\sim}\right) \right)}{\rho \varepsilon S \pi^2 (1 + 2 k\sim)^2 v}$$

Осуществим теперь предельный переход:

> Ck1 := Limit(Ck1, epsilon=0); Ck1 := factor(value(Ck1));

$$Ck1 := \lim_{\varepsilon \rightarrow 0} \frac{4 P \sqrt{2} l\sim^{(3/2)} (-1)^{k\sim} \left(\sin\left(\frac{\pi \varepsilon}{2 l\sim}\right) \cos\left(\frac{\pi \varepsilon k\sim}{l\sim}\right) + \cos\left(\frac{\pi \varepsilon}{2 l\sim}\right) \sin\left(\frac{\pi \varepsilon k\sim}{l\sim}\right) \right)}{\rho \varepsilon S \pi^2 (1 + 2 k\sim)^2 v}$$

$$Ck1 := \frac{2 \sqrt{l\sim} P \sqrt{2} (-1)^{k\sim}}{\pi (1 + 2 k\sim) \rho S v}$$

Рассмотрим второй вариант задания начальных условий — с дельта-функцией. В этом случае будем иметь формулу для коэффициентов Фурье:

> Ck2 := Int(phi2(x) * ef(k, x), x=0..l) / ev(k)^(1/2) / v;

$$Ck2 := \frac{\sqrt{4}}{\sqrt{\frac{\pi^2 (1 + 2 k\sim)^2}{l\sim^2}} v} \int_0^{l\sim} \frac{P \text{Dirac}(x - l\sim) \sin\left(\frac{\pi (1 + 2 k\sim) x}{2 l\sim}\right) \sqrt{2}}{\rho S \sqrt{l\sim}} dx$$

> assume(epsilon>0): Ck2 := simplify(value(Ck2));

$$Ck2 := \frac{2 \sqrt{l\sim} P \sqrt{2} (-1)^{k\sim}}{\pi (1 + 2 k\sim) \rho S v}$$

Результат, как и следовало, один и тот же! Определим окончательно коэффициенты разложения

> C := unapply(Ck2, k);

$$C := k\sim \rightarrow \frac{2 \sqrt{l\sim} P \sqrt{2} (-1)^{k\sim}}{\pi (1 + 2 k\sim) \rho S v}$$

Итак, мы получили формальное решение задачи:

> sol := spr;

$$sol := \sum_{k\sim=1}^{\infty} \left(\frac{4 P (-1)^{k\sim} \sin\left(\frac{\pi v (1 + 2 k\sim) t}{2 l\sim}\right) \sin\left(\frac{\pi (1 + 2 k\sim) x}{2 l\sim}\right)}{\pi (1 + 2 k\sim) \rho S v} \right)$$

Это решение можно преобразовать, введя в рассмотрение массу стержня $M = \rho S l$:

```
> subs(rho=M/l/S, op(1, sol));
```

$$\frac{4 P l (-1)^{k-1} \sin\left(\frac{\pi v (1 + 2 k-1) t}{2 l}\right) \sin\left(\frac{\pi (1 + 2 k-1) x}{2 l}\right)}{\pi (1 + 2 k-1) M v}$$

```
> solution:=Sum(%,k=1..infinity);
```

$$solution := \sum_{k=1}^{\infty} \left(\frac{4 P l (-1)^{k-1} \sin\left(\frac{\pi v (1 + 2 k-1) t}{2 l}\right) \sin\left(\frac{\pi (1 + 2 k-1) x}{2 l}\right)}{\pi (1 + 2 k-1) M v} \right)$$

Таким образом, мы окончательно получили решение задачи в виде

$$u(x, t) = \frac{4Pl}{\pi v M} \sum_{k=1}^{\infty} \frac{(-1)^k \sin\left(\frac{\pi(2k+1)}{2l} vt\right) \sin\left(\frac{\pi(2k+1)}{2l} x\right)}{(2k+1)}.$$

Задача 4.3. Найти функцию, гармоническую внутри круга радиуса R и такую, что $u|_{r=R} = f(\varphi)$, где:

1. $f(\varphi) = \cos^2(\varphi)$, $R = 1$;
2. $f(\varphi) = \sin^3(\varphi)$, $R = 1$;
3. $f(\varphi) = \cos^6(\varphi) + \sin^6(\varphi)$, $R = 1$;
4. $f(\varphi) = \cos(3\varphi)$, $R = 1$.

Решение. Воспользуемся готовым решением задачи Дирихле для круга и выполним решения наших задач в Maple. Для этого разработаем процедуру, реализующую известное решение рассматриваемой задачи в рядах [3]:

```
> Dirichlet:=proc(f,R)
local a,b;
a:=n->1/Pi*Int(f*cos(n*phi),phi=-Pi..Pi);
b:=n->1/Pi*Int(f*sin(n*phi),phi=-Pi..Pi);
a(0)/2+
add(r^n/R^n*(a(n)*cos(n*phi)+b(n)*sin(n*phi)),
n=1..Order);
RETURN(map(simplify,value(%)));end proc;
```

Эта процедура-функция строит отрезок ряда до порядка `Order` или точное решение в виде сравнительно простой формулы, если это возможно.

Применим нашу программу для решения поставленных задач. Вариант первый:

```
> f:=cos(phi)^2;R:=1;
```

$$f := \cos(\phi)^2$$

$$R := 1$$

```
> sol:=Dirichlet(f,R);
```

$$sol := \frac{1}{2} + \frac{1}{2} r^2 \cos(2\phi)$$

Проверим найденное решение:

```
> linalg[laplacian](sol,[r,phi],coords=polar);
```

```
simplify(subs(r=R,sol)-f);
```

0

0

Вариант второй:

```
> f:=sin(phi)^3;R:=1;
```

$$f := \sin(\phi)^3$$

$$R := 1$$

```
> sol:=Dirichlet(f,R);
```

$$sol := \frac{3}{4} r \sin(\phi) - \frac{1}{4} r^3 \sin(3\phi)$$

Проверка:

```
> combine(linalg[laplacian](sol,[r,phi],coords=polar));
```

```
simplify(subs(r=R,sol)-f);
```

0

0

Вариант третий:

```
> f:=sin(phi)^6+cos(phi)^6;R:=1;
```

```
sol:=Dirichlet(f,R);
```

$$f := \sin(\phi)^6 + \cos(\phi)^6$$

$$R := 1$$

$$sol := \frac{5}{8} + \frac{3}{8} r^4 \cos(4\phi)$$

Проверка:

```
> linalg[laplacian](sol,[r,phi],coords=polar);
```

```
simplify(subs(r=R,sol)-f);
```

0
0

Вариант четвертый:

```
> f:=cos(3*phi);R:=1;  
sol:=Dirichlet(f,R);
```

$$f := \cos(3\phi)$$
$$R := 1$$
$$sol := r^3 \cos(3\phi)$$

Проверка:

```
> linalg[laplacian](sol,[r,phi],coords=polar);  
simplify(subs(r=R,sol)-f);  
0  
0
```

Задача 4.4. Решить смешанную краевую задачу:

$$\frac{\partial^2 u}{\partial t^2} - \frac{\partial^2 u}{\partial x^2} + 4u = 0, \quad 0 < x < 1,$$

$$u|_{x=0} = u|_{x=1} = 0, \quad u|_{t=0} = x^2 - x, \quad \frac{\partial u}{\partial t}|_{t=0} = 0.$$

Решение. Сформулируем нашу задачу в Maple:

```
> restart;with(linalg):  
Warning, the protected names norm and trace have been  
redefined and unprotected  
> eq:=diff(u(x,t),t,t)-diff(u(x,t),x,x)+4*u(x,t)=0;  
eq :=  $\left(\frac{\partial^2}{\partial t^2} u(x,t)\right) - \left(\frac{\partial^2}{\partial x^2} u(x,t)\right) + 4 u(x,t) = 0$   
> init_c:=u(x,0)=x^2-x,D[2](u)(x,0)=0;  
init_c :=  $u(x,0) = x^2 - x, D_2(u)(x,0) = 0$   
> bound_c:=u(0,t) = 0,u(1,t)=0;  
bound_c :=  $u(0,t) = 0, u(1,t) = 0$   
> ic1:=rhs(init_c[1]);  
ic1 :=  $x^2 - x$ 
```

Разделяем переменные

```

> pde_sol:=pdsolve(eq,HINT=X(x)*T(t));
      pde_sol := (u(x, t) = X(x) T(t)) &where
      [ {  $\frac{d^2}{dx^2} X(x) = -c_1 X(x), \frac{d^2}{dt^2} T(t) = -c_1 T(t) - 4 T(t) \} ]$ 
> eq1:=op(1,op(1,op(2,pde_sol)));
      eq2:=op(2,op(1,op(2,pde_sol)));

      eq1 :=  $\frac{d^2}{dx^2} X(x) = -c_1 X(x)$ 
      eq2 :=  $\frac{d^2}{dt^2} T(t) = -c_1 T(t) - 4 T(t)$ 

```

В результате мы получили два обыкновенных дифференциальных уравнения. Сформулируем теперь соответствующую задачу Штурма-Лиувилля

```

> eq1:=subs(_c[1]=-lambda,eq1);s1:=X(0)=0; s2:=X(1)=0;

      eq1 :=  $\frac{d^2}{dx^2} X(x) = -\lambda X(x)$ 
      s1 := X(0) = 0
      s2 := X(1) = 0

```

Находим общее решение уравнения Штурма-Лиувилля

```

> assume(lambda>0):res1:=dsolve(eq1,X(x));
      res1 := X(x) = _C1 sin( $\sqrt{\lambda} x$ ) + _C2 cos( $\sqrt{\lambda} x$ )
> X:=unapply( rhs(res1),x);
      X := x → _C1 sin( $\sqrt{\lambda} x$ ) + _C2 cos( $\sqrt{\lambda} x$ )

```

Используя однородные граничные условия, получим характеристическое уравнение для определения собственных значений задачи Штурма-Лиувилля

```

> sist:={s1,s2};
      sist := {_C2 = 0, _C1 sin( $\sqrt{\lambda}$ ) + _C2 cos( $\sqrt{\lambda}$ ) = 0}
> linalg[genmatrix](sist, {_C1, _C2});
      [ 0      1
      sin( $\sqrt{\lambda}$ )  cos( $\sqrt{\lambda}$ ) ]
> Delta:=linalg[det](%);
      Δ := -sin( $\sqrt{\lambda}$ )

```

Находим собственные значения (решаем уравнение $-\Delta = \sin \sqrt{\lambda} = 0$)

```

> _EnvAllSolutions := true:solve(Delta=0,lambda);

```

$$\pi^2 _Z1 \sim^2$$

Итак, собственные значения образуют счетное множество чисел $\lambda = \lambda_k = \pi^2 k^2$, $k = 1, 2, \dots$. Определим их в Maple как функцию натурального аргумента

```
> subs(_Z1='k', %):ev:=unapply(%, k);
      ev := k → π² k²
```

Найдем теперь соответствующие собственные функции

```
> X:='X':assume(k, posint);
> eq1:=subs(lambda=ev(k), eq1);
```

$$eq1 := \frac{d^2}{dx^2} X(x) = -\pi^2 k^2 X(x)$$

```
> dsolve({eq1, s1, s2}, X(x));
```

$$X(x) = _C1 \sin(\pi k x)$$

Напомним, что собственные функции определяются с точностью до постоянного множителя. Сразу же нормируем собственные функции

```
> rhs(%) / sqrt(int(rhs(%)^2, x=0..1));
```

$$\frac{_C1 \sin(\pi k x) \sqrt{2}}{\sqrt{_C1^2}}$$

```
> simplify(% , radical, symbolic);
ef:=unapply(%, (k, x));
```

$$ef := (k, x) \rightarrow \sin(\pi k x) \sqrt{2}$$

Итак, мы получили собственные значения и соответствующие им нормированные собственные функции

```
> ev(k); ef(k, x);
```

$$\pi^2 k^2$$

$$\sin(\pi k x) \sqrt{2}$$

Решаем теперь второе дифференциальное уравнение (по переменной t)

```
> eq2:=subs(_c[1]=-ev(k), eq2);
```

$$eq2 := \frac{d^2}{dt^2} T(t) = -\pi^2 k^2 T(t) - 4 T(t)$$

```
> res2:=dsolve({eq2, D(T)(0)=0}, T(t));
```

$$res2 := T(t) = _C2 \cos(\sqrt{\pi^2 k^2 + 4} t)$$

При решении этого уравнения мы сразу учли одно из начальных условий (однородное).

Таким образом, решение нашей задачи мы ищем в форме ряда (принцип суперпозиции)

```
> spr:=Sum(C(k)*op(2,rhs(res2))*ef(k,x),k=1..infinity);
```

$$spr := \sum_{k=1}^{\infty} C(k) \cos(\sqrt{\pi^2 k^2 + 4} t) \sin(\pi k x) \sqrt{2}$$

Находим коэффициенты разложения

```
> Ck:=int(ic1*ef(k,x),x=0..1);
```

$$Ck := \frac{2\sqrt{2}((-1)^k - 1)}{\pi^3 k^3}$$

```
> C:=unapply(Ck,k);
```

$$C := k \rightarrow \frac{2\sqrt{2}((-1)^k - 1)}{\pi^3 k^3}$$

Следовательно, решение задачи дается формулой

```
> sol:=spr;
```

$$sol := \sum_{k=1}^{\infty} \left(\frac{4((-1)^k - 1) \cos(\sqrt{\pi^2 k^2 + 4} t) \sin(\pi k x)}{\pi^3 k^3} \right)$$

Удобно преобразовать это решение к другому виду. Индекс k заменим на $2m+1$:

```
> subs(k=2*m+1,op(1,sol));assume(m,integer);
```

$$\frac{4((-1)^{(2m+1)} - 1) \cos(\sqrt{\pi^2 (2m+1)^2 + 4} t) \sin(\pi (2m+1) x)}{\pi^3 (2m+1)^3}$$

```
> simplify(%):factor(%);
```

$$-\frac{8 \cos(\sqrt{4\pi^2 m^2 + 4\pi^2 m + \pi^2 + 4} t) \sin(\pi (2m+1) x)}{\pi^3 (2m+1)^3}$$

Окончательно мы получаем

```
> 'u(x,t)'=Sum(%,m=0..infinity);
```

$$u(x,t) = \sum_{m=0}^{\infty} \left(-\frac{8 \cos(\sqrt{4\pi^2 m^2 + 4\pi^2 m + \pi^2 + 4} t) \sin(\pi (2m+1) x)}{\pi^3 (2m+1)^3} \right)$$

То есть решение задачи определяется по формуле

$$u(x,t) = -\frac{8}{\pi^3} \sum_{m=0}^{\infty} \frac{\sin((2m+1)\pi x)}{(2m+1)^3} \cos\left(\sqrt{(2m+1)^2 \pi^2 + 4} t\right).$$

Задача 4.5. Решить смешанную краевую задачу:

$$\frac{\partial^2 u}{\partial t^2} + 2 \frac{\partial u}{\partial t} - \frac{\partial^2 u}{\partial x^2} + u = 0, \quad 0 < x < \pi,$$

$$u|_{x=0} = u|_{x=\pi} = 0, \quad u|_{t=0} = \pi x - x^2, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = 0.$$

Решение. Сформулируем задачу в Maple:

```
> restart;with(linalg):
Warning, the protected names norm and trace have been
redefined and unprotected
> eq:=
diff(u(x,t),t,t)+2*diff(u(x,t),t)-
diff(u(x,t),x,x)+u(x,t)=0;
eq := ( (d^2 u(x,t) / dt^2) + 2 (d u(x,t) / dt) - (d^2 u(x,t) / dx^2) + u(x,t) = 0
> init_c:=u(x,0)=Pi*x-x^2,D[2](u)(x,0)=0;
init_c := u(x,0) = pi x - x^2, D2(u)(x,0) = 0
> bound_c:=u(0,t) = 0,u(Pi,t)=0;
bound_c := u(0,t) = 0, u(pi,t) = 0
> ic1:=rhs(init_c[1]);
ic1 := pi x - x^2
```

Разделяем переменные и получаем два обыкновенных дифференциальных уравнения

```
> pde_sol:=pdsolve(eq,HINT=X(x)*T(t));
pde_sol := (u(x,t) = X(x) T(t)) &where
[ { d^2 X(x) / dx^2 = -c1 X(x), d^2 T(t) / dt^2 = -2 (d T(t) / dt) + -c1 T(t) - T(t) } ]
> eq1:=op(1,op(1,op(2,pde_sol)));
eq2:=op(2,op(1,op(2,pde_sol)));
eq1 := d^2 X(x) / dx^2 = -c1 X(x)
eq2 := d^2 T(t) / dt^2 = -2 (d T(t) / dt) + -c1 T(t) - T(t)
```

Формулируем соответствующую задачу Штурма-Лиувилля

```
> eq1:=subs(_c[1]=-lambda,eq1);s1:=X(0)=0; s2:=X(Pi)=0;
```

$$eq1 := \frac{d^2}{dx^2} X(x) = -\lambda X(x)$$

$$s1 := X(0) = 0$$

$$s2 := X(\pi) = 0$$

Решаем эту задачу

```
> assume(lambda>0):res1:=dsolve(eq1,X(x));
```

$$res1 := X(x) = _C1 \sin(\sqrt{\lambda} x) + _C2 \cos(\sqrt{\lambda} x)$$

```
> X:=unapply(rhs(res1),x);
```

$$X := x \rightarrow _C1 \sin(\sqrt{\lambda} x) + _C2 \cos(\sqrt{\lambda} x)$$

```
> sist:={s1,s2};
```

$$sist := \{ _C2 = 0, _C1 \sin(\sqrt{\lambda} \pi) + _C2 \cos(\sqrt{\lambda} \pi) = 0 \}$$

```
> linalg[genmatrix](sist,[_C1,_C2]);
```

$$\begin{bmatrix} 0 & 1 \\ \sin(\sqrt{\lambda} \pi) & \cos(\sqrt{\lambda} \pi) \end{bmatrix}$$

```
> Delta:=linalg[det](%);
```

$$\Delta := -\sin(\sqrt{\lambda} \pi)$$

```
> _EnvAllSolutions := true:solve(Delta=0,lambda);
```

$$_Z1 \sim^2$$

```
> subs(_Z1='k',%):ev:=unapply(%,k);
```

$$ev := k \rightarrow k^2$$

Мы получили счетное множество собственных значений: $\lambda = \lambda_k = k^2$, $k = 1, 2, 3, \dots$

Находим теперь соответствующие собственные функции:

```
> X:='X':assume(k, posint);
```

```
> eq1:=subs(lambda=ev(k),eq1);
```

$$eq1 := \frac{d^2}{dx^2} X(x) = -k^2 X(x)$$

```
> dsolve({eq1,s1,s2},X(x));
```

$$X(x) = _C1 \sin(k x)$$

Сразу же нормируем эти функции

```
> rhs(%)/sqrt(int(rhs(%)^2,x=0..Pi));
```

$$\frac{C1 \sin(k x) \sqrt{2}}{\sqrt{\pi} _C1^2}$$

```
> simplify(% , radical , symbolic);
```

> ef:=unapply(%,(k,x));

$$\frac{\sin(kx)\sqrt{2}}{\sqrt{\pi}}$$

$$ef := (k, x) \rightarrow \frac{\sin(kx)\sqrt{2}}{\sqrt{\pi}}$$

Итак, собственные значения и нормированные собственные функции будут

> ev(k);ef(k,x);

$$k^{-2}$$

$$\frac{\sin(kx)\sqrt{2}}{\sqrt{\pi}}$$

Решаем второе дифференциальное уравнение по переменной t

> eq2:=subs(_c[1]=-ev(k),eq2);

$$eq2 := \frac{d^2}{dt^2} T(t) = -2 \left(\frac{d}{dt} T(t) \right) - k^2 T(t) - T(t)$$

> res2:=dsolve({eq2,D(T)(0)=0},T(t));

$$res2 := T(t) = _C1 e^{(-t)} \sin(k t) + _C1 e^{(-t)} \cos(k t) k$$

> factor(rhs(%));

$$_C1 e^{(-t)} (\sin(k t) + \cos(k t) k)$$

> %/_C1;

$$e^{(-t)} (\sin(k t) + \cos(k t) k)$$

> Tk:=unapply(%,t,k);

$$Tk := (t, k) \rightarrow e^{(-t)} (\sin(t k) + \cos(t k) k)$$

Таким образом, воспользовавшись принципом суперпозиции, составляем формальный ряд

$$u(x, t) = \sqrt{\frac{2}{\pi}} \sum_{k=1}^{\infty} C_k e^{-t} (\sin kt + k \cos kt) \sin kx.$$

Определяем теперь коэффициенты разложения

> Ck:=int(ic1*ef(k,x),x=0..Pi);

$$Ck := -\frac{2\sqrt{2}((-1)^k - 1)}{k^3\sqrt{\pi}}$$

Удобно преобразовать эти коэффициенты, заменив k на $2m+1$

> assume(m, posint);

simplify(subs(k=2*m+1,Ck));

$$\frac{4\sqrt{2}}{(2m+1)^3\sqrt{\pi}}$$

> C:=unapply(%,m);

$$C := m \rightarrow \frac{4\sqrt{2}}{(2m+1)^3\sqrt{\pi}}$$

Таким образом, решение задачи

> sol:=Sum(C(m)*Tk(t,2*m+1)*ef(2*m+1,x),m=0..infinity);

$$sol := \sum_{m=0}^{\infty} \left(\frac{8e^{-t} (\sin(t(2m+1)) + \cos(t(2m+1))(2m+1)) \sin((2m+1)x)}{(2m+1)^3\pi} \right)$$

Итак, окончательно мы получили

$$u(x,t) = \frac{8e^{-t}}{\pi} \sum_{m=0}^{\infty} \frac{[\sin(2m+1)t + (2m+1)\cos(2m+1)t] \sin(2m+1)x}{(2m+1)^3}.$$

Задача 4.6. Струна $0 \leq x \leq l$ с жестко закрепленными концами до момента $t = 0$ находилась в состоянии равновесия под действием поперечной силы $F_0 = const$, приложенной к точке x_0 струны перпендикулярно к невозмущенному положению струны. В начальный момент времени $t = 0$ действие силы F_0 мгновенно прекращается. Найти колебания струны при $t > 0$.

Решение. Сформулируем математическую задачу. Требуется найти функцию $u(x,t)$, удовлетворяющую одномерному волновому уравнению

$$\frac{1}{v^2} \frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}, \quad 0 < x < l, \quad t > 0,$$

и граничным условиям

$$u|_{x=0} = 0, \quad u|_{x=l} = 0.$$

Кроме того, функция $u(x,t)$ удовлетворяет начальным условиям, одно из которых,

$$\left. \frac{\partial u}{\partial t} \right|_{t=0} = 0,$$

выражает равенство нулю начальной скорости точек струны. Найдем начальное отклонение струны (рис. 4.1); для этого достаточно определить величину h .




Рис. 4.1. К задаче 4.6

Из условия равновесия (в проекциях на вертикальную ось) находим

$$T(\sin \alpha + \sin \beta) = F_0.$$

В силу малости отклонений $\sin \alpha \approx \operatorname{tg} \alpha$, $\sin \beta \approx \operatorname{tg} \beta$ ⁵, но

$$\operatorname{tg} \alpha = \frac{h}{x_0}, \quad \operatorname{tg} \beta = \frac{h}{l - x_0}.$$

Таким образом,

$$h = \frac{F_0 x_0 (l - x_0)}{lT}.$$

Итак, второе начальное условие будет таким

$$u|_{t=0} = \begin{cases} \frac{h}{x_0} x, & 0 < x \leq x_0, \\ \frac{h(l-x)}{l-x_0}, & x_0 \leq x < l. \end{cases}$$

Решаем сформулированную задачу в Maple. Определяем уравнение и дополнительные условия

```
> eq:=diff(u(x,t),t,t)-v^2*diff(u(x,t),x,x)=0;
```

$$eq := \left(\frac{\partial^2}{\partial t^2} u(x, t) \right) - v^2 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) = 0$$

```
> ic1:=u(x,0)=phi(x);
```

```
ic2:=D[2](u)(x,0)=psi(x);
```

⁵ В силу малости отклонения, T не зависит от отклонения.

```

ic1 := u(x, 0) = phi(x)
ic2 := D2(u)(x, 0) = psi(x)
> bound_c := u(0, t) = 0, u(l, t) = 0;
      bound_c := u(0, t) = 0, u(l, t) = 0
> phi := piecewise(x > 0 and x < x0, h*x/x0, x > x0 and x < l,
      h*(1-x)/(1-x0));
      phi := {
      { h*x / x0, -x < 0 and x - x0 < 0
      { h*(1-x) / (1-x0), -x + x0 < 0 and x - l < 0
> psi := 0;
      psi := 0

```

Разделяем переменные в волновом уравнении и формулируем задачу Штурма-Лиувилля

```

> pde_sol := pdsolve(eq, HINT=X(x)*T(t));
eq1 := op(1, op(1, op(2, pde_sol)));
eq2 := op(2, op(1, op(2, pde_sol)));
pde_sol := (u(x, t) = X(x) T(t)) &where
      [ { d^2/dx^2 X(x) = -c1 X(x), d^2/dt^2 T(t) = v^2 -c1 T(t) } ]
      eq1 := d^2/dx^2 X(x) = -c1 X(x)
      eq2 := d^2/dt^2 T(t) = v^2 -c1 T(t)
> eq1 := subs(_c[1] = -lambda, eq1); s1 := X(0) = 0; s2 := X(l) = 0;
      eq1 := d^2/dx^2 X(x) = -lambda X(x)
      s1 := X(0) = 0
      s2 := X(l) = 0

```

Находим собственные значения задачи

```

> assume(lambda > 0); res1 := dsolve(eq1, X(x));
      res1 := X(x) = _C1 sin(sqrt(lambda) x) + _C2 cos(sqrt(lambda) x)
> X := unapply(rhs(%), x);
      X := x -> _C1 sin(sqrt(lambda) x) + _C2 cos(sqrt(lambda) x)

```

```

> sist := {s1, s2};
      sist := {_C2 = 0, _C1 sin(sqrt(lambda) l) + _C2 cos(sqrt(lambda) l) = 0}
> linalg[genmatrix](sist, {_C1, _C2});
      [ 0          1
      sin(sqrt(lambda) l)  cos(sqrt(lambda) l) ]
> char := linalg[det](%) = 0;
      char := -sin(sqrt(lambda) l) = 0
> _EnvAllSolutions := true;
      solve(char, lambda);
      
$$\frac{\pi^2 \_Z1^2}{l^2}$$


```

Итак, собственные значения будут

$$\lambda = \lambda_k = \frac{\pi^2 k^2}{l^2}, \quad k = 1, 2, 3, \dots$$

Определяем их в Maple как функцию целочисленного аргумента

```

> subs(_Z1='k', %);
      ev := unapply(%, k);
      
$$ev := k \rightarrow \frac{\pi^2 k^2}{l^2}$$


```

Находим теперь собственные функции

```

> X := 'X': assume(k, posint);
      subs(lambda=ev(k), eq1);
      
$$\frac{d^2}{dx^2} X(x) = -\frac{\pi^2 k^2}{l^2} X(x)$$

> dsolve({%, s1, s2}, X(x));
      
$$X(x) = \_C1 \sin\left(\frac{\pi k x}{l}\right)$$


```

Сразу же нормируем эти функции

```

> rhs(%) / sqrt(int(rhs(%)^2, x=0..l));
      
$$\frac{\_C1 \sin\left(\frac{\pi k x}{l}\right) \sqrt{2}}{\sqrt{\_C1^2 l}}$$


```

Итак, нормированные собственные функции будут

$$X_k(x) = \frac{\sqrt{2}}{l} \sin\left(\frac{k\pi x}{l}\right),$$

определим их в Maple

```
> simplify(% , radical , symbolic );
ef := unapply(% , (k , x) );
```

$$\frac{\sin\left(\frac{\pi k x}{l}\right) \sqrt{2}}{\sqrt{l}}$$

$$ef := (k, x) \rightarrow \frac{\sin\left(\frac{\pi k x}{l}\right) \sqrt{2}}{\sqrt{l}}$$

Решение задачи ищем в виде ряда

```
> sol := Sum(T[k](t) * ef(k, x), k=1..infinity);
```

$$sol := \sum_{k=1}^{\infty} \frac{T_k(t) \sin\left(\frac{\pi k x}{l}\right) \sqrt{2}}{\sqrt{l}}$$

Для коэффициентов разложения, очевидно, будем иметь следующую задачу

```
> ode := {diff(T[k](t), t, t) + v^2 * ev(k) * T[k](t) = 0,
T[k](0) = simplify(int(op(2, phi) * ef(k, x), x=0..x0) +
int(op(4, phi) * ef(k, x), x=x0..l)),
D(T[k])(0) = int(psi(x) * ef(k, x), x=0..l)};
```

$$ode := \left\{ \left(\frac{d^2}{dt^2} T_{k-}(t) \right) + \frac{v^2 \pi^2 k^{-2} T_{k-}(t)}{l^2} = 0, D(T_{k-})(0) = 0, \right.$$

$$\left. T_{k-}(0) = \frac{l^{(5/2)} h \sqrt{2} \sin\left(\frac{x0 \pi k-}{l}\right)}{x0 \pi^2 k^{-2} (l - x0)} \right\}$$

Решаем эту задачу

```
> simplify(dsolve(ode, T[k](t)));
```

$$T_{k-}(t) = \frac{l^{(5/2)} h \sqrt{2} \sin\left(\frac{x0 \pi k-}{l}\right) \cos\left(\frac{\pi k- v t}{l}\right)}{x0 \pi^2 k^{-2} (l - x0)}$$

и окончательно получаем

```
> sol := subs(% , sol);
```

$$sol := \sum_{k=1}^{\infty} \left(\frac{2 l^2 h \sin\left(\frac{x_0 \pi k}{l}\right) \cos\left(\frac{\pi k v t}{l}\right) \sin\left(\frac{\pi k x}{l}\right)}{x_0 \pi^2 k^2 (l - x_0)} \right)$$

Итак, мы нашли решение в виде формального ряда

$$u(x, t) = \frac{2l^2 h}{\pi^2 x_0 (l - x_0)} \sum_{k=1}^{\infty} \frac{1}{k^2} \sin\left(\frac{k\pi x_0}{l}\right) \sin\left(\frac{k\pi x}{l}\right) \cos\left(\frac{k\pi v t}{l}\right),$$

где h определяется по формуле

$$h = \frac{F_0 x_0 (l - x_0)}{lT}.$$

Исключив h , окончательно получим

$$u(x, t) = \frac{2lF_0}{\pi^2 T} \sum_{k=1}^{\infty} \frac{1}{k^2} \sin\left(\frac{k\pi x_0}{l}\right) \sin\left(\frac{k\pi x}{l}\right) \cos\left(\frac{k\pi v t}{l}\right).$$

Задача 4.7. Изучить осесимметричные колебания круглой мембраны радиуса a , вызванные ударным импульсом P , приложенным в момент времени $t = 0$ и распределенным по площади круга радиуса ε .

Решение. Сформулируем математическую задачу. Требуется найти функцию $u(r, t)$, удовлетворяющую уравнению осесимметричных колебаний мембраны

$$\frac{1}{v^2} \frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial r^2} + \frac{1}{r} \frac{\partial u}{\partial r} + \frac{1}{r^2} \frac{\partial^2 u}{\partial \varphi^2}$$

при краевом условии $u|_{r=a} = 0$ (мембрана закреплена по контуру) и начальных условиях $u|_{t=0} = 0$,

$$\left. \frac{\partial u}{\partial t} \right|_{t=0} = \begin{cases} \frac{P}{\pi \varepsilon^2 \rho}, & 0 \leq r < \varepsilon, \\ 0, & \varepsilon < r \leq a. \end{cases}$$

Воспользуемся для решения задачи системой Maple. Определим задачу:

```
> eq:=diff(u(r,t),t$2)-
v^2*VectorCalculus[Laplacian](u(r,t),'polar'[r,theta])=0;
eq := \left( \frac{\partial^2}{\partial t^2} u(r, t) \right) - \frac{v^2 \left( \left( \frac{\partial}{\partial r} u(r, t) \right) + r \left( \frac{\partial^2}{\partial r^2} u(r, t) \right) \right)}{r} = 0
```

```
> ic1:=u(r,0)=0;
ic2:=D[2](u)(x,0)=f(r);
ic1 := u(r, 0) = 0
```

```

ic2 := D2(u)(x, 0) = f(r)
> bound_c := u(a, t) = 0;
      bound_c := u(a, t) = 0
> f := piecewise(r > 0 and r < epsilon, P/Pi/epsilon^2/rho,
      r > epsilon and r < a, 0);
      f := {
              P / (pi * epsilon^2 * rho)   -r < 0 and r - epsilon < 0
              0                             epsilon - r < 0 and r - a < 0
            }

```

Здесь мы воспользовались еще одним полезным оператором — оператором-лапласианом `Laplacian` из пакета `VectorCalculus`.

Разделяем переменные и получаем два обыкновенных дифференциальных уравнения

```

> pde_sol := pdsolve(eq, HINT=R(r)*T(t));
> eq1 := op(1, op(1, op(2, pde_sol)));
eq2 := op(2, op(1, op(2, pde_sol)));
pde_sol := (u(r, t) = R(r) T(t)) &where
          [ [ {
              {
                d^2 R(r) = R(r) _c1 - d R(r) / dr,
                d^2 T(t) = v^2 T(t) _c1
              }
            } ] ]
          eq1 := d^2 R(r) = R(r) _c1 - d R(r) / dr
          eq2 := d^2 T(t) = v^2 T(t) _c1

```

Формулируем соответствующую задачу Штурма-Лиувилля

```

> eq1 := subs(_c[1] = -lambda, eq1); s1 := R(a) = 0;
          eq1 := d^2 R(r) = -R(r) lambda - d R(r) / dr
          s1 := R(a) = 0

```

Находим теперь общее решение уравнения Штурма-Лиувилля:

```

> res1 := dsolve(eq1, R(r));
      res1 := R(r) = _C1 BesselJ(0, sqrt(lambda) r) + _C2 BesselY(0, sqrt(lambda) r)

```

Итак, мы получили

$$R(r) = C_1 J_0(\sqrt{\lambda}r) + C_2 Y_0(\sqrt{\lambda}r),$$

где $J_0(x)$, $Y_0(x)$ — функции Бесселя первого и второго рода с нулевым индексом. Как известно (см., например, [3]), только функция Бесселя $J_0(x)$ первого рода ограничена в нуле. Поэтому, чтобы получить ограниченное решение, мы должны в общем решении положить $C_2 = 0$. Тогда собственные функции будут $R_n(r) = J_0\left(\frac{\alpha_n r}{a}\right)$, где α_n — n -ый положительный корень уравнения $J_0(x) = 0$.

Определим найденные собственные функции в Maple и сразу же вычислим квадрат нормы этих функций

```
> R:=unapply(subs({_C2=0, _C1=1, sqrt(lambda)=alpha[n]/a},
rhs(%)),n,r);
```

$$R := (n, r) \rightarrow \text{BesselJ}\left(0, \frac{\alpha_n r}{a}\right)$$

```
> norm2:=int(r*R(n,r)^2,r=0..a);
```

$$\text{norm2} := \frac{1}{2} \frac{a^2 (\sqrt{\pi} \alpha_n \text{BesselJ}(0, \alpha_n)^2 + \sqrt{\pi} \alpha_n \text{BesselJ}(1, \alpha_n)^2)}{\sqrt{\pi} \alpha_n}$$

```
> norm2:=simplify(norm2, {BesselJ(0, alpha[n])=0});
```

$$\text{norm2} := \frac{1}{2} a^2 \text{BesselJ}(1, \alpha_n)^2$$

Решение задачи ищем в форме ряда

```
> sol:=Sum(T[n](t)*R(n,r),n=1..infinity);
```

$$\text{sol} := \sum_{n=1}^{\infty} T_n(t) \text{BesselJ}\left(0, \frac{\alpha_n r}{a}\right)$$

Рассматриваем теперь второе, полученное при разделении переменных, обыкновенное дифференциальное уравнение для функции $T(t)$:

```
> eq2:=subs({T(t)=T[n](t), _c[1]=-(alpha[n]/a)^2}, eq2);
```

$$\text{eq2} := \frac{d^2}{dt^2} T_n(t) = -\frac{v^2 T_n(t) \alpha_n^2}{a^2}$$

Находим решение этого уравнения с учетом однородного начального условия

```
> ode_sol:=simplify(dsolve({eq2, T[n](0)=0}, T[n](t)));
```

$$\text{ode_sol} := T_n(t) = _C1 \sin\left(\frac{v \alpha_n t}{a}\right)$$

Тогда решение нашей задачи будет:

```
> sol:=subs(subs(_C1=C[n], ode_sol), sol);
```

$$sol := \sum_{n=1}^{\infty} C_n \sin\left(\frac{v \alpha_n t}{a}\right) \text{BesselJ}\left(0, \frac{\alpha_n r}{a}\right)$$

Остается только определить коэффициенты полученного разложения. Для этого воспользуемся вторым неоднородным начальным условием

> Diff(u(r,t),t)=diff(sol,t);

$$\frac{\partial}{\partial t} u(r,t) = \sum_{n=1}^{\infty} \frac{C_n \cos\left(\frac{v \alpha_n t}{a}\right) v \alpha_n \text{BesselJ}\left(0, \frac{\alpha_n r}{a}\right)}{a}$$

> 'f(r)'=value(subs(t=0,rhs(%)));

$$f(r) = \sum_{n=1}^{\infty} \frac{C_n v \alpha_n \text{BesselJ}\left(0, \frac{\alpha_n r}{a}\right)}{a}$$

Пользуясь соответствующей теоремой разложения [3], находим

> simplify(op(1,rhs(%))/BesselJ(0,alpha[n]*r/a)=
Int(r*op(2,f)*R(n,r),r=0..epsilon)/norm2;

$$\frac{C_n v \alpha_n}{a} = \frac{2}{a^2 \text{BesselJ}(1, \alpha_n)^2} \int_0^{\epsilon} \frac{r P \text{BesselJ}\left(0, \frac{\alpha_n r}{a}\right)}{\pi \epsilon^2 \rho} dr$$

Откуда находим коэффициенты разложения

> isolate(value(%),C[n]);

$$C_n = \frac{2 P \text{BesselJ}\left(1, \frac{\epsilon \alpha_n}{a}\right)}{\pi \epsilon \rho \alpha_n^2 \text{BesselJ}(1, \alpha_n)^2 v}$$

Таким образом, будем иметь

> 'u(r,t)'=subs(% ,sol);

$$u(r,t) = \sum_{n=1}^{\infty} \left(\frac{2 P \text{BesselJ}\left(1, \frac{\epsilon \alpha_n}{a}\right) \sin\left(\frac{v \alpha_n t}{a}\right) \text{BesselJ}\left(0, \frac{\alpha_n r}{a}\right)}{\pi \epsilon \rho \alpha_n^2 \text{BesselJ}(1, \alpha_n)^2 v} \right)$$

Итак, окончательно мы получили решение в виде ряда

$$u(r,t) = \frac{2P}{v\pi\epsilon\rho} \sum_{n=1}^{\infty} \frac{J_1\left(\frac{\alpha_n \epsilon}{a}\right)}{\alpha_n^2 J_1^2(\alpha_n)} J_0\left(\frac{\alpha_n r}{a}\right) \sin\left(\frac{\alpha_n vt}{a}\right).$$

Задача 4.8. Решить следующую смешанную задачу

$$\frac{\partial u}{\partial t} = x \frac{\partial^2 u}{\partial x^2} + \frac{\partial u}{\partial x} - \frac{9}{4x} u, \quad 0 < x < 1,$$

$$|u|_{x=0} < \infty, \quad u|_{x=1} = 0, \quad u|_{t=0} = J_3(\mu_k \sqrt{x}),$$

где μ_k — положительный корень уравнения $J_3(\mu) = 0$.

Решение. Определяем задачу в Maple

```
> restart;
> eq:=diff(u(x,t),t)=
x*diff(u(x,t),x,x)+diff(u(x,t),x)-9/4/x*u(x,t);
eq :=  $\frac{\partial}{\partial t} u(x, t) = x \left( \frac{\partial^2}{\partial x^2} u(x, t) \right) + \left( \frac{\partial}{\partial x} u(x, t) \right) - \frac{9}{4} \frac{u(x, t)}{x}$ 
> ic:=u(x,0)=BesselJ(3,mu[k]*x^(1/2));
ic :=  $u(x, 0) = \text{BesselJ}(3, \mu_k \sqrt{x})$ 
> bc:=u(1,t) = 0;
bc :=  $u(1, t) = 0$ 
```

Выполняем разделение переменных и получаем два обыкновенных дифференциальных уравнения

```
> pde_sol:=pdsolve(eq,HINT=X(x)*T(t));
eq1:=op(1,op(1,op(2,pde_sol)));
eq2:=op(2,op(1,op(2,pde_sol)));
pde_sol := (u(x, t) = X(x) T(t)) &where [

$$\left\{ \frac{d}{dt} T(t) = \frac{1}{4} T(t) - c_1, \frac{d^2}{dx^2} X(x) = \frac{1}{4} \frac{X(x) - c_1}{x} - \frac{1}{4} \frac{4 \left( \frac{d}{dx} X(x) \right) x - 9 X(x)}{x^2} \right\}$$

eq1 :=  $\frac{d^2}{dx^2} X(x) = \frac{1}{4} \frac{X(x) - c_1}{x} - \frac{1}{4} \frac{4 \left( \frac{d}{dx} X(x) \right) x - 9 X(x)}{x^2}$ 
eq2 :=  $\frac{d}{dt} T(t) = \frac{1}{4} T(t) - c_1$ 
```

Формулируем соответствующую задачу Штурма-Лиувилля

```
> eq1:=subs(_c[1]=-lambda,eq1);s1:=X(1)=0;
```

$$eq1 := \frac{d^2}{dx^2} X(x) = -\frac{1}{4} \frac{X(x) \lambda}{x} - \frac{1}{4} \frac{4 \left(\frac{d}{dx} X(x) \right) x - 9 X(x)}{x^2}$$

$$s1 := X(1) = 0$$

Находим общее решение уравнения Штурма-Лиувилля

```
> res1 := dsolve(eq1, X(x));
res1 := X(x) = _C1 BesselJ(3, sqrt(lambda) sqrt(x)) + _C2 BesselY(3, sqrt(lambda) sqrt(x))
```

Учитывая требование ограниченности искомой функции в нуле, сразу же находим совокупность собственных значений и собственных функций. Очевидно, в общем решении мы должны положить $_C2 = 0$ (ограниченной в нуле является только функция Бесселя первого рода) и, таким образом, собственные функции будут $X_k(x) = J_3(\mu_k \sqrt{x})$, где μ_k — положительные корни уравнения $J_3(\mu) = 0$, а собственные значения $\lambda_k = \mu_k^2$, $k = 1, 2, \dots$

Как видим, наше уравнение Штурма-Лиувилля есть уравнение типа Бесселя. Действительно, с помощью соответствующей замены переменных, данное уравнение сводится к уравнению Бесселя с индексом 3. Покажем это с помощью Maple. Прежде всего, преобразуем уравнение к удобочитаемому виду

```
> collect(DEtools[DEnormal](eq1, x, X(x)), X(x));
(lambda x - 9) X(x) + 4 (d/dx X(x)) x + 4 x^2 (d^2/dx^2 X(x)) = 0
```

Здесь мы воспользовались полезной командой `DEnormal` из пакета `DEtools`. Эта команда возвращает нормализованную (упорядоченную) форму дифференциального уравнения. Формат вызова команды:

```
DEnormal(des, ivar, dvar)
```

где `des` — обыкновенное дифференциальное уравнение, `ivar` — независимая переменная, `dvar` — зависимая переменная.

Следующие замены переменных приводят это уравнение к классическому уравнению Бесселя

```
> tr := {x=y/lambda, X(x)=Y(y)};
itr := simplify(solve(tr, {y, Y(y)}));
go := PDEtools[dchange](tr, eq1, [y, Y(y)]);
DEtools[DEnormal](go, y, Y(y));
dsolve(%, Y(y));
```

$$tr := \left\{ x = \frac{y}{\lambda}, X(x) = Y(y) \right\}$$

$$itr := \{ y = \lambda x, Y(y) = X(x) \}$$

$$go := \lambda^2 \left(\frac{d^2}{dy^2} Y(y) \right) = \frac{1}{4} \frac{Y(y) \lambda^2}{y} - \frac{1}{4} \frac{\left(4 \left(\frac{d}{dy} Y(y) \right) y - 9 Y(y) \right) \lambda^2}{y^2}$$

$$Y(y) y - 9 Y(y) + 4 \left(\frac{d}{dy} Y(y) \right) y + 4 y^2 \left(\frac{d^2}{dy^2} Y(y) \right) = 0$$

$$Y(y) = _C1 \text{ BesselJ}(3, \sqrt{y}) + _C2 \text{ BesselY}(3, \sqrt{y})$$

```
> tr1 := {y=z^2, Y(y)=Z(z)};
go1 := PDEtools[dchange](tr1, go, [z, Z(z)]);
DEtools[DEnormal](go1, z, Z(z));
dsolve(%, Z(z));
```

$$tr1 := \{y = z^2, Y(y) = Z(z)\}$$

$$go1 := \frac{1}{2} \lambda^2 \frac{\left(\frac{1}{2} \frac{d}{dz} Z(z) \frac{1}{z^2} + \frac{1}{2} \frac{d^2}{dz^2} Z(z) \frac{1}{z} \right)}{z} =$$

$$\frac{1}{4} \frac{Z(z) \lambda^2}{z^2} - \frac{1}{4} \frac{\left(2 z \left(\frac{d}{dz} Z(z) \right) - 9 Z(z) \right) \lambda^2}{z^4}$$

$$Z(z) z^2 - 9 Z(z) + z \left(\frac{d}{dz} Z(z) \right) + z^2 \left(\frac{d^2}{dz^2} Z(z) \right) = 0$$

$$Z(z) = _C1 \text{ BesselJ}(3, z) + _C2 \text{ BesselY}(3, z)$$

Здесь мы сразу же получили и общие решения соответствующих уравнений. Замены переменных выполнены с помощью команды `dchange` из пакета `PDEtools`, о которой мы уже кратко говорили в главе 3.

Вернемся к нашей задаче. Определим найденные собственные функции в Maple

```
> R := unapply(
subs({_C2=0, _C1=1, sqrt(lambda)=mu[k]}, rhs(res1)), k, x);
R := (k, x) -> BesselJ(3, mu[k] sqrt(x))
```

Проверим, что найденные собственные функции ортогональны на отрезке $[0, 1]$. При упрощении интеграла нам придется воспользоваться характеристическим уравнением $J_3(\mu) = 0$. Поэтому подготовим соответствующие соотношения, вытекающие из этого уравнения, которые и будем использовать при преобразовании интеграла

```
> simplify(BesselJ(3, mu[k]))=0;
e1:=isolate(%, BesselJ(0, mu[k]));
e1n:=subs(k=n, e1); e1m:=subs(k=m, e1);
```


$$\frac{-8 \operatorname{BesselJ}(1, \mu_k) + 4 \operatorname{BesselJ}(0, \mu_k) \mu_k + \operatorname{BesselJ}(1, \mu_k) \mu_k^2}{\mu_k^2} = 0$$

$$e1 := \operatorname{BesselJ}(0, \mu_k) = \frac{1}{4} \frac{8 \operatorname{BesselJ}(1, \mu_k) - \operatorname{BesselJ}(1, \mu_k) \mu_k^2}{\mu_k}$$

$$e1n := \operatorname{BesselJ}(0, \mu_n) = \frac{1}{4} \frac{8 \operatorname{BesselJ}(1, \mu_n) - \operatorname{BesselJ}(1, \mu_n) \mu_n^2}{\mu_n}$$

$$e1m := \operatorname{BesselJ}(0, \mu_m) = \frac{1}{4} \frac{8 \operatorname{BesselJ}(1, \mu_m) - \operatorname{BesselJ}(1, \mu_m) \mu_m^2}{\mu_m}$$

Теперь вычисляем интеграл

```
> Int(R(m, x) * R(n, x), x=0..1);
simplify(subs({e1n, e1m}, value(%)));
```

$$\int_0^1 \operatorname{BesselJ}(3, \mu_m \sqrt{x}) \operatorname{BesselJ}(3, \mu_n \sqrt{x}) dx$$

0

Как видим, этот интеграл равен нулю, что и означает ортогональность найденных собственных функций.

При разложении произвольной функции в ряд по найденным собственным функциям понадобится квадрат нормы этих функций. Вычислим его

```
> norm2 := Int(R(k, x)^2, x=0..1);
simplify(subs({e1}, value(%)));
```

$$norm2 := \int_0^1 \operatorname{BesselJ}(3, \mu_k \sqrt{x})^2 dx$$

$$\frac{1}{16} \operatorname{BesselJ}(1, \mu_k)^2 \mu_k^2$$

Итак, мы получили формулу

$$\int_0^1 J_3(\mu_m \sqrt{x}) J_3(\mu_n \sqrt{x}) = \begin{cases} 0, & m \neq n, \\ \frac{\mu_n^2}{16} J_1^2(\mu_n), & m = n. \end{cases}$$

Решение исходной задачи ищем в форме ряда

```
> sol := Sum(T[n](t) * R(n, x), n=1..infinity);
```

$$sol := \sum_{n=1}^{\infty} T_n(t) \text{BesselJ}(3, \mu_n \sqrt{x})$$

Для коэффициентов разложения имеем второе дифференциальное уравнение, полученное при разделении переменных

$$> \text{eq2} := \text{subs}(\{T(t) = T[n](t), _c[1] = -\mu[n]^2\}, \text{eq2});$$

$$\text{eq2} := \frac{d}{dt} T_n(t) = -\frac{1}{4} T_n(t) \mu_n^2$$

Общее решение этого уравнения

$$> \text{ode_sol} := \text{simplify}(\text{dsolve}(\text{eq2}, T[n](t)));$$

$$\text{ode_sol} := T_n(t) = _C1 e^{\left(-1/4 \mu_n^2 t\right)}$$

Таким образом, решение задачи определяется следующим рядом

$$> \text{sol} := \text{subs}(\text{subs}(_C1 = C[n], \text{ode_sol}), \text{sol});$$

$$\text{sol} := \sum_{n=1}^{\infty} C_n e^{\left(-1/4 \mu_n^2 t\right)} \text{BesselJ}(3, \mu_n \sqrt{x})$$

Из начального условия будем иметь

$$> R(k, x) = \text{value}(\text{subs}(t=0, \text{sol}));$$

$$\text{BesselJ}(3, \mu_k \sqrt{x}) = \sum_{n=1}^{\infty} C_n \text{BesselJ}(3, \mu_n \sqrt{x})$$

Откуда, учитывая свойство ортогональности собственных функций, находим $C_n = 0$, если $n \neq k$, и $C_k = 1$. Таким образом, решение задачи дается формулой

$$u(x, t) = e^{-\frac{\mu_k^2 t}{4}} J_3(\mu_k \sqrt{x}).$$

Задача 4.9. Решить следующую смешанную задачу

$$\frac{\partial^2 u}{\partial t^2} = x \frac{\partial^2 u}{\partial x^2} + \frac{\partial u}{\partial x} - \frac{1}{x} u, \quad 0 < x < 1,$$

$$\left| u \right|_{x=0} < \infty, \quad u|_{x=1} = 0, \quad u|_{t=0} = 0, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = J_2(\mu_k \sqrt{x}),$$

где μ_k — положительный корень уравнения $J_2(\mu) = 0$.

Решение. Определяем задачу в Maple

$$> \text{restart};$$

$$> \text{eq} := \text{diff}(u(x, t), t, t) =$$

```

x*diff(u(x,t),x,x)+diff(u(x,t),x)-1/x*u(x,t);
eq :=  $\frac{\partial^2}{\partial t^2} u(x,t) = x \left( \frac{\partial^2}{\partial x^2} u(x,t) \right) + \left( \frac{\partial}{\partial x} u(x,t) \right) - \frac{u(x,t)}{x}$ 
> ic1:=u(x,0)=0;ic2:=BesselJ(2,mu[k]*x^(1/2));
ic1 := u(x, 0) = 0
ic2 := D2(u)(x, 0) = BesselJ(2, μk√x)
> bc:=u(1,t) = 0;
bc := u(1, t) = 0

```

Выполняем разделение переменных и получаем два обыкновенных дифференциальных уравнения

```

> pde_sol:=pdsolve(eq,HINT=X(x)*T(t));
> eq1:=op([2,1,1],pde_sol);eq2:=op([2,1,2],pde_sol);
pde_sol := (u(x, t) = X(x) T(t)) &where

$$\left[ \left\{ \frac{d^2}{dx^2} X(x) = \frac{X(x) \_c_1}{x} - \frac{\left( \frac{d}{dx} X(x) \right) x - X(x)}{x^2}, \frac{d^2}{dt^2} T(t) = T(t) \_c_1 \right\} \right]$$

eq1 :=  $\frac{d^2}{dx^2} X(x) = \frac{X(x) \_c_1}{x} - \frac{\left( \frac{d}{dx} X(x) \right) x - X(x)}{x^2}$ 
eq2 :=  $\frac{d^2}{dt^2} T(t) = T(t) \_c_1$ 

```

Формулируем соответствующую задачу Штурма-Лиувилля

```

> eq1:=subs(_c[1]=-lambda,eq1);s1:=X(1)=0;
eq1 :=  $\frac{d^2}{dx^2} X(x) = -\frac{X(x) \lambda}{x} - \frac{\left( \frac{d}{dx} X(x) \right) x - X(x)}{x^2}$ 
s1 := X(1) = 0

```

Находим общее решение уравнения Штурма-Лиувилля

```

> res1:=dsolve(eq1,X(x));
res1 := X(x) = _C1 BesselJ(2, 2√λ√x) + _C2 BesselY(2, 2√λ√x)

```

Учитывая требование ограниченности искомой функции в нуле, находим совокупность собственных значений и собственных функций. В общем решении мы полагаем $_C2 = 0$ (ограниченной в нуле является только функция Бесселя первого рода) и, таким образом, собственные функции будут $X_k(x) = J_2(\mu_k \sqrt{x})$, где μ_k —

положительные корни уравнения $J_2(\mu) = 0$, а собственные значения $\lambda_k = \frac{\mu_k^2}{4}$, $k = 1, 2, \dots$.

Определим найденные собственные функции в Maple

```
> R:=unapply(subs(_C2=0, _C1=1,
  2*sqrt(lambda)*sqrt(x)=mu[k]*sqrt(x), rhs(%)), k, x);
  R := (k, x) -> BesselJ(2, mu_k*sqrt(x))
```

Проверим, что найденные собственные функции ортогональны на отрезке $[0, 1]$. При упрощении интеграла нам придется воспользоваться характеристическим уравнением $J_2(x) = 0$. Поэтому подготовим соответствующие соотношения, вытекающие из этого уравнения, которые и будем использовать при преобразовании интеграла

```
> simplify(BesselJ(2, mu[k]))=0;
  -2 BesselJ(1, mu_k) + BesselJ(0, mu_k) mu_k = 0
  mu_k

> e1:=isolate(%, BesselJ(0, mu[k]));
  e1n:=subs(k=n, e1); e1m:=subs(k=m, e1);

  e1 := BesselJ(0, mu_k) = 2 BesselJ(1, mu_k) / mu_k
  e1n := BesselJ(0, mu_n) = 2 BesselJ(1, mu_n) / mu_n
  e1m := BesselJ(0, mu_m) = 2 BesselJ(1, mu_m) / mu_m
```

Теперь вычисляем интеграл

```
> Int(R(m, x)*R(n, x), x=0..1);
  simplify(subs({e1n, e1m}, value(%)));
  \int_0^1 BesselJ(2, mu_m*sqrt(x)) BesselJ(2, mu_n*sqrt(x)) dx
```

0

Как видим, этот интеграл равен нулю — это и означает ортогональность найденных собственных функций.

При разложении произвольной функции в ряд по найденным собственным функциям понадобится квадрат нормы этих функций. Вычислим его

```
> norm2 := Int(R(k, x)^2, x=0..1);
simplify(subs({e1}, value(%)));
```

$$norm2 := \int_0^1 \text{BesselJ}(2, \mu_k \sqrt{x})^2 dx$$

$$\text{BesselJ}(1, \mu_k)^2$$

Итак, мы получили формулу

$$\int_0^1 J_2(\mu_m \sqrt{x}) J_2(\mu_n \sqrt{x}) dx = \begin{cases} 0, & m \neq n, \\ J_1^2(\mu_n), & m = n. \end{cases}$$

Решение исходной задачи ищем в форме ряда

```
> sol := Sum(T[n](t) * R(n, x), n=1..infinity);
```

$$sol := \sum_{n=1}^{\infty} T_n(t) \text{BesselJ}(2, \mu_n \sqrt{x})$$

Для определения коэффициентов разложения воспользуемся вторым дифференциальным уравнением, полученным при разделении переменных

```
> eq2 := subs({T(t)=T[n](t), _c[1]=-mu[n]^2/4}, eq2);
```

$$eq2 := \frac{d^2}{dt^2} T_n(t) = -\frac{1}{4} T_n(t) \mu_n^2$$

Общее решение этого уравнения с учетом однородного начального условия будет

```
> ic_1 := subs(u(x,0)=T[n](0), ic1);
```

```
ode_sol := simplify(dsolve({eq2, ic_1}, T[n](t)));
```

$$ic_1 := T_n(0) = 0$$

$$ode_sol := T_n(t) = _C1 \sin\left(\frac{1}{2} \mu_n t\right)$$

Таким образом, решение задачи определяется следующим рядом

```
> sol := subs(subs(_C1=C[n], ode_sol), sol);
```

$$sol := \sum_{n=1}^{\infty} C_n \sin\left(\frac{1}{2} \mu_n t\right) \text{BesselJ}(2, \mu_n \sqrt{x})$$

Из второго начального условия будем иметь

```
> R(k, x) = value(subs(t=0, diff(sol, t)));
```

$$\text{BesselJ}(2, \mu_k \sqrt{x}) = \sum_{n=1}^{\infty} \left(\frac{1}{2} C_n \mu_n \text{BesselJ}(2, \mu_n \sqrt{x}) \right)$$

Откуда, учитывая свойство ортогональности собственных функций, находим $C_n = 0$, если $n \neq k$, и $\frac{1}{2}C_k\mu_k = 1 \Rightarrow C_k = \frac{2}{\mu_k}$. Таким образом, решение задачи дается формулой

$$u(x, t) = \frac{2}{\mu_k} \sin\left(\frac{\mu_k t}{2}\right) J_2(\mu_k \sqrt{x}).$$

Задача 4.10. Решить следующую смешанную задачу

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + \frac{1}{x} \frac{\partial u}{\partial x} - \frac{9}{x^2} u, \quad 0 < x < 1,$$

$$\left|u\right|_{x=0} < \infty, \quad u|_{x=1} = 0, \quad u|_{t=0} = J_3(\mu_1 x), \quad \left.\frac{\partial u}{\partial t}\right|_{t=0} = J_3(\mu_1 x),$$

где μ_1 — положительный корень уравнения $J_3(\mu) = 0$.

Решение. Определяем задачу в Maple

```
> restart;
> eq:=diff(u(x,t),t,t)=
diff(u(x,t),x,x)+1/x*diff(u(x,t),x)-9/x^2*u(x,t);
eq :=  $\frac{\partial^2}{\partial t^2} u(x, t) = \left(\frac{\partial^2}{\partial x^2} u(x, t)\right) + \frac{\frac{\partial}{\partial x} u(x, t)}{x} - \frac{9 u(x, t)}{x^2}$ 
> ic1:=u(x,0)= BesselJ(3,mu[1]*x);ic2:=BesselJ(3,mu[1]*x);
ic1 := u(x, 0) = BesselJ(3,  $\mu_1 x$ )
ic2 := D2(u)(x, 0) = BesselJ(3,  $\mu_1 x$ )
> bc:=u(1,t) = 0;
bc := u(1, t) = 0
```

Выполним разделение переменных и получим два обыкновенных дифференциальных уравнения

```
> pde_sol:=pdsolve(eq,HINT=X(x)*T(t));
> eq1:=op([2,1,1],pde_sol);eq2:=op([2,1,2],pde_sol);
pde_sol := (u(x, t) = X(x) T(t)) &where

$$\left[ \left\{ \frac{d^2}{dx^2} X(x) = X(x) \_c_1 - \frac{\left(\frac{d}{dx} X(x)\right)x - 9 X(x)}{x^2}, \frac{d^2}{dt^2} T(t) = T(t) \_c_1 \right\} \right]$$

```

$$eq1 := \frac{d^2}{dx^2} X(x) = X(x) _c1 - \frac{\left(\frac{d}{dx} X(x)\right) x - 9 X(x)}{x^2}$$

$$eq2 := \frac{d^2}{dt^2} T(t) = T(t) _c1$$

Формулируем соответствующую задачу Штурма-Лиувилля

> eq1:=subs(_c[1]=-lambda, eq1); s1:=X(1)=0;

$$eq1 := \frac{d^2}{dx^2} X(x) = -X(x) \lambda - \frac{\left(\frac{d}{dx} X(x)\right) x - 9 X(x)}{x^2}$$

$$s1 := X(1) = 0$$

Находим общее решение уравнения Штурма-Лиувилля

> collect(DEtools[DEnormal](eq1, x, X(x)), X(x));

$$(\lambda x^2 - 9) X(x) + \left(\frac{d}{dx} X(x)\right) x + \left(\frac{d^2}{dx^2} X(x)\right) x^2 = 0$$

> res1:=dsolve(eq1, X(x));

$$res1 := X(x) = _C1 \text{BesselJ}(3, \sqrt{\lambda} x) + _C2 \text{BesselY}(3, \sqrt{\lambda} x)$$

С учетом требования ограниченности искомой функции в нуле, находим совокупность собственных значений и собственных функций. Полагаем $_C2 = 0$ и, таким образом, собственные функции будут $X_k(x) = J_3(\mu_k x)$, где μ_k — положительные корни уравнения $J_3(\mu) = 0$, а собственные значения $\lambda_k = \mu_k^2$, $k = 1, 2, \dots$

Определим найденные собственные функции в Maple

> R:=unapply(
subs(_C2=0, _C1=1, sqrt(lambda)=mu[k], rhs(%)), k, x);
R := (k, x) → BesselJ(3, $\mu_k x$)

Найденные собственные функции должны быть ортогональны на отрезке $[0, 1]$ с весом x . Убедимся в этом. При упрощении интеграла нам придется воспользоваться характеристическим уравнением $J_3(\mu) = 0$. Поэтому подготовим соответствующие соотношения, вытекающие из этого уравнения, которые и будем использовать при преобразовании интеграла

> simplify(BesselJ(3, mu[k]))=0:
> e1:=isolate(%, BesselJ(1, mu[k]));
e1n:=subs(k=n, e1); e1m:=subs(k=m, e1);

$$e1 := \text{BesselJ}(1, \mu_k) = -\frac{4 \text{BesselJ}(0, \mu_k) \mu_k}{-8 + \mu_k^2}$$

$$e1n := \text{BesselJ}(1, \mu_n) = -\frac{4 \text{BesselJ}(0, \mu_n) \mu_n}{-8 + \mu_n^2}$$

$$e1m := \text{BesselJ}(1, \mu_m) = -\frac{4 \text{BesselJ}(0, \mu_m) \mu_m}{-8 + \mu_m^2}$$

Теперь вычисляем интеграл

> Int(x*R(n,x)*R(m,x),x=0..1);

$$\int_0^1 x \text{BesselJ}(3, \mu_n x) \text{BesselJ}(3, \mu_m x) dx$$

> simplify(subs({e1m,e1n},value(%)));

0

Интеграл равен нулю — это и означает ортогональность найденных собственных функций на отрезке $[0,1]$ с весом x .

Вычислим квадрат нормы собственных функций

> norm2:=Int(x*R(k,x)^2,x=0..1);

simplify(subs({e1},value(%)));

$$norm2 := \int_0^1 x \text{BesselJ}(3, \mu_k x)^2 dx$$

$$\frac{1}{2} \frac{\text{BesselJ}(0, \mu_k)^2 \mu_k^4}{(-8 + \mu_k^2)^2}$$

Итак, мы получили формулу

$$\int_0^1 x J_3(\mu_m x) J_3(\mu_n x) = \begin{cases} 0, & m \neq n, \\ \frac{\mu_n^4 J_0^2(\mu_n)}{2(\mu_n^2 - 8)^2}, & m = n. \end{cases}$$

Решение исходной задачи ищем в форме ряда

> sol:=Sum(T[n](t)*R(n,x),n=1..infinity);

$$sol := \sum_{n=1}^{\infty} T_n(t) \text{BesselJ}(3, \mu_n x)$$

Коэффициенты этого разложения удовлетворяют второму дифференциальному уравнению, полученному при разделении переменных

> eq2:=subs({T(t)=T[n](t),_c[1]=-mu[n]^2},eq2);

$$eq2 := \frac{d^2}{dt^2} T_n(t) = -T_n(t) \mu_n^2$$

Находим общее решение этого уравнения

> ode_sol:=simplify(dsolve(eq2,T[n](t)));

$$ode_sol := T_n(t) = _C1 \sin(\mu_n t) + _C2 \cos(\mu_n t)$$

Таким образом, решение задачи определяется следующим рядом

> sol:=subs(subs({_C1=A[n],_C2=B[n]},ode_sol),sol);

$$sol := \sum_{n=1}^{\infty} (A_n \sin(\mu_n t) + B_n \cos(\mu_n t)) \text{BesselJ}(3, \mu_n x)$$

Из начальных условий будем иметь

> rhs(ic1)=value(subs(t=0,sol));

rhs(ic2)=value(subs(t=0,diff(sol,t)));

$$\text{BesselJ}(3, \mu_1 x) = \sum_{n=1}^{\infty} B_n \text{BesselJ}(3, \mu_n x)$$

$$\text{BesselJ}(3, \mu_1 x) = \sum_{n=1}^{\infty} A_n \mu_n \text{BesselJ}(3, \mu_n x)$$

Откуда, учитывая свойство ортогональности собственных функций, находим $B_n = 0$

и $A_n = 0$, если $n \neq 1$, и $B_1 = 1$, $A_1 \mu_1 = 1 \Rightarrow A_1 = \frac{1}{\mu_1}$. Таким образом, решение задачи

дается формулой

$$u(x,t) = \left[\frac{1}{\mu_1} \sin(\mu_1 t) + \cos(\mu_1 t) \right] J_3(\mu_1 x).$$

Упражнения

1. Концы струны закреплены жестко, а начальное отклонение имеет форму квадратичной параболы, симметричной относительно перпендикуляра к середине струны. Найти колебания струны, если начальные скорости равны нулю, а начальное максимальное смещение равно h .

Ответ. $u(x,t) = \frac{32h}{\pi^3} \sum_{k=0}^{\infty} \frac{1}{(2k+1)^3} \sin\left(\frac{(2k+1)\pi x}{l}\right) \cos\left(\frac{(2k+1)\pi vt}{l}\right).$

2. Струна с жестко закрепленными концами возбуждается ударом жесткого плоского молоточка, сообщаящего ей следующее начальное распределение скоростей:

$$\left. \frac{\partial u}{\partial t} \right|_{t=0} = \psi(x) = \begin{cases} 0, & 0 \leq x \leq x_0 - \delta, \\ v_0, & x_0 - \delta \leq x \leq x_0 + \delta, \\ 0, & x_0 + \delta \leq x \leq l. \end{cases}$$

Найти колебания струны, если начальное отклонение равно нулю.

$$\text{Ответ. } u(x, t) = \frac{4v_0 l}{\pi^2 v} \sum_{k=1}^{\infty} \frac{1}{k^2} \sin\left(\frac{k\pi x_0}{l}\right) \sin\left(\frac{k\pi\delta}{l}\right) \sin\left(\frac{k\pi x}{l}\right) \sin\left(\frac{k\pi v t}{l}\right).$$

3. Струна с жестко закрепленными концами возбуждается ударом жесткого выпуклого молоточка, сообщаящего ей следующее начальное распределение скоростей:

$$\left. \frac{\partial u}{\partial t} \right|_{t=0} = \psi(x) = \begin{cases} 0, & 0 \leq x \leq x_0 - \delta, \\ v_0 \cos\left(\frac{\pi(x-x_0)}{2\delta}\right), & x_0 - \delta \leq x \leq x_0 + \delta, \\ 0, & x_0 + \delta \leq x \leq l. \end{cases}$$

Найти колебания струны, если начальное отклонение равно нулю.

$$\text{Ответ. } u(x, t) = \frac{8v_0\delta}{\pi^2 v} \sum_{k=1}^{\infty} \frac{1}{k} \frac{\sin\left(\frac{k\pi x_0}{l}\right) \cos\left(\frac{k\pi\delta}{l}\right)}{1 - \left(\frac{2k\delta}{l}\right)^2} \sin\left(\frac{k\pi x}{l}\right) \sin\left(\frac{k\pi v t}{l}\right).$$

4. Струна с жестко закрепленными концами возбуждается ударом острого молоточка, передающего ей импульс I в точке x_0 . Найти колебания струны, если начальное отклонение равно нулю.

$$\text{Ответ. } u(x, t) = \frac{2I}{\pi v \rho} \sum_{k=1}^{\infty} \frac{1}{k} \sin\left(\frac{k\pi x_0}{l}\right) \sin\left(\frac{k\pi x}{l}\right) \sin\left(\frac{k\pi v t}{l}\right), \text{ где } \rho \text{ — линейная}$$

плотность массы струны.

5. Найти продольные колебания стержня, один конец которого ($x = 0$) закреплен жестко, а другой ($x = l$) свободен, при начальных условиях:

$$u(x, 0) = kx, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = 0, \quad 0 \leq x \leq l.$$

$$\text{Ответ. } u(x, t) = \frac{8kl}{\pi^2} \sum_{n=0}^{\infty} \frac{(-1)^n}{(2n+1)^3} \sin\left(\frac{(2n+1)\pi x}{2l}\right) \cos\left(\frac{(2n+1)\pi v t}{2l}\right).$$

6. Стержень длины l , закрепленный в точке $(x = 0)$, растянут силой $F_0 = const$, приложенной на его другом конце. В момент $t = 0$ действие силы F_0 мгновенно прекращается. Найти колебания стержня, если начальные скорости равны нулю.

Ответ. Ответ получается из ответа предыдущей задачи 5, если положить $k = F_0/(ES)$, где E — модуль упругости, а S — площадь поперечного сечения стержня.

7. Найти колебания упругого стержня со свободными концами, если начальные скорости и начальные смещения в продольном направлении произвольны.

Ответ. Решением краевой задачи

$$\begin{aligned} u_{tt} &= v^2 u_{xx}, \quad 0 < x < l, \quad t > 0, \\ u(x, 0) &= \varphi(x), \quad u_t(x, 0) = \psi(x), \quad 0 < x < l, \\ u_x(0, t) &= u_x(l, t) = 0, \quad t > 0 \end{aligned}$$

является:

$$\begin{aligned} u(x, t) &= \frac{1}{l} \int_0^l [\varphi(\xi) + t\psi(\xi)] d\xi + \sum_{k=1}^{\infty} \left[a_k \cos\left(\frac{k\pi vt}{l}\right) + b_k \sin\left(\frac{k\pi vt}{l}\right) \right] \cos\left(\frac{k\pi x}{l}\right), \\ a_k &= \frac{1}{l} \int_0^l \varphi(\xi) \cos \frac{k\pi \xi}{l} d\xi, \quad b_k = \frac{1}{l} \int_0^l \psi(\xi) \sin \frac{k\pi \xi}{l} d\xi. \end{aligned}$$

8. Найти колебания упругого стержня со свободными концами, получившего в начальный момент времени продольный импульс I в один из концов.

Ответ.
$$u(x, t) = -\frac{I}{\rho l} t - \frac{2I}{\pi v \rho} \sum_{k=1}^{\infty} \frac{(-1)^k}{k} \sin\left(\frac{k\pi vt}{l}\right) \cos\left(\frac{k\pi x}{l}\right).$$

Указание. Решение задачи может быть получено из решения предыдущей задачи 7, если положить $\varphi(x) = 0$,

$$\psi(x) = \begin{cases} 0, & 0 \leq x \leq l - \delta, \\ -\frac{I}{\delta \rho}, & l - \delta < x \leq l, \end{cases}$$

а затем перейти к пределу при $\delta \rightarrow 0$, либо положить $\varphi(x) = 0$, а

$$\psi(x) = -\frac{I}{\rho} \delta(x - x_0), \quad 0 < x_0 < l,$$

где $\delta(x)$ — дельта-функция Дирака, а затем перейти к пределу при $x_0 \rightarrow 0$.

Решая задачу в системе Maple можно просто считать, что $x_0 = l$ (см. решение задачи 4.2).

9. Найти колебания упругого стержня, один конец которого жестко закреплен, а другой конец свободен и получает в начальный момент времени продольный импульс I .

$$\text{Ответ. } u(x, t) = -\frac{4I}{\pi v \rho} \sum_{k=0}^{\infty} \frac{(-1)^k}{2k+1} \sin\left(\frac{(2k+1)\pi vt}{2l}\right) \cos\left(\frac{(2k+1)\pi x}{2l}\right).$$

10. Найти распределение температуры в стержне $0 \leq x \leq l$ с теплоизолированной боковой поверхностью, если температура его концов поддерживается равной нулю, а начальная температура равна $T_0 = \text{const}$.

$$\text{Ответ. } u(x, t) = \frac{4T_0}{\pi} \sum_{k=0}^{\infty} \frac{1}{2k+1} e^{-\frac{(2k+1)^2 \pi^2 v^2}{l^2} t} \sin\left(\frac{(2k+1)\pi x}{2l}\right).$$

11. Дать решение первой краевой задачи для уравнения Лапласа вне круга.

$$\text{Ответ. } u(r, \varphi) = \frac{A_0}{2} + \sum_{n=1}^{\infty} \left(\frac{a}{r}\right)^n (A_n \cos n\varphi + B_n \sin n\varphi),$$

где $A_n = \frac{1}{\pi} \int_0^{2\pi} f(\varphi) \cos n\varphi d\varphi$, $n = 0, 1, 2, \dots$, $B_n = \frac{1}{\pi} \int_0^{2\pi} f(\varphi) \sin n\varphi d\varphi$, $n = 1, 2, \dots$, a — радиус круга.

12. Дать решение второй краевой задачи для уравнения Лапласа вне круга.

$$\text{Ответ. } u(r, \varphi) = \sum_{n=1}^{\infty} \frac{a^{n+1}}{nr^n} (A_n \cos n\varphi + B_n \sin n\varphi) + C,$$

где A_n и B_n коэффициенты Фурье функции $f(\varphi) = \frac{\partial u}{\partial r} \Big|_{r=a}$; a — радиус круга; C — произвольная постоянная.

13. Дать решение второй краевой задачи для уравнения Лапласа внутри круга.

Ответ. $u(r, \varphi) = \sum_{n=1}^{\infty} \frac{r^n}{na^{n-1}} (A_n \cos n\varphi + B_n \sin n\varphi) + C$, где использованы те же обозначения, что и в задаче 12.

14. Дать решение третьей внутренней краевой задачи для уравнения Лапласа в круге, если граничное условие записывается в виде

$$\frac{\partial u}{\partial r} - hu \Big|_{r=a} = -f.$$

$$\text{Ответ. } u(r, \varphi) = \sum_{n=1}^{\infty} \frac{r^n}{(n-ah)a^{n-1}} (A_n \cos n\varphi + B_n \sin n\varphi) - \frac{A_0}{2h},$$

где A_n и B_n коэффициенты Фурье функции $f(\varphi)$, определяемые по формулам задачи 11; a — радиус круга.

15. Дать решение третьей внешней краевой задачи для уравнения Лапласа для круга (граничное условие задачи 14).

$$\text{Ответ. } u(r, \varphi) = -\sum_{n=1}^{\infty} \frac{a^n}{(an+h)r^n} (A_n \cos n\varphi + B_n \sin n\varphi) - \frac{A_0}{2h}, \text{ где использованы те}$$

же обозначения, что и в задаче 14.

16. Дать решение внутренней краевой задачи для уравнения Лапласа, если на границе круга задано условие

$$u|_{r=a} = A \sin \varphi.$$

$$\text{Ответ. } u = \frac{Ar}{a} \sin \varphi.$$

17. Дать решение внутренней краевой задачи для уравнения Лапласа, если на границе круга задано условие

$$u|_{r=a} = A \sin^3 \varphi + B.$$

$$\text{Ответ. } u = B + \frac{3Ar}{a} \sin \varphi - 4A \left(\frac{r}{a}\right)^3 \sin 3\varphi.$$

18. Найти функцию, гармоническую внутри круга радиуса R с центром в начале координат и такую, что

$$\text{а) } \left. \frac{\partial u}{\partial r} \right|_{r=R} = A \cos(\varphi);$$

$$\text{б) } \left. \frac{\partial u}{\partial r} \right|_{r=R} = A \cos(2\varphi);$$

$$\text{в) } \left. \frac{\partial u}{\partial r} \right|_{r=R} = \sin^3(\varphi).$$

$$\text{Ответ. а) } u(r, \varphi) = Ar \cos \varphi + C; \text{ б) } u(r, \varphi) = \frac{A}{2R} r^2 \cos 2\varphi + C;$$

$$\text{в) } u(r, \varphi) = \frac{1}{4} \left(3r \sin \varphi - \frac{r^2}{3R^2} \sin 3\varphi \right) + C; \text{ где } C \text{ — произвольная постоянная.}$$

19. Найти стационарное распределение температуры $u(r, \varphi)$ внутри бесконечного цилиндра радиуса R , если на его поверхности поддерживается температура

$$u|_{r=R} = A \sin \varphi.$$

Ответ. $u = \frac{Ar}{R} \sin \varphi.$

20. Найти функцию, гармоническую внутри единичного круга с центром в начале координат и такую, что $u|_{r=1} = \cos^4(\varphi)$. *Указание.* См. решение задачи 4.3.

21. Решить задачу о колебаниях струны $0 < x < l$ с закрепленными концами, если начальные скорости точек струны равны нулю, а начальное отклонение имеет форму синусоиды

$$u|_{t=0} = A \sin\left(\frac{\pi n x}{l}\right), \quad (n \text{ — целое}).$$

Ответ. $u(x, t) = A \sin\left(\frac{\pi n x}{l}\right) \cos\left(\frac{\pi n v t}{l}\right).$

22. Найти закон выравнивания заданного начального распределения температуры $f(x)$ в пластине, грани которой $x = 0$ и $x = a$ не пропускают тепла.

Ответ. $u(x, t) = \frac{a_0}{2} + \sum_{k=1}^{\infty} a_k e^{-\frac{v^2 \pi^2 k^2 t}{l^2}} \cos\left(\frac{k \pi x}{l}\right), \quad a_k = \frac{2}{l} \int_0^l f(\xi) \cos\left(\frac{k \pi \xi}{l}\right) d\xi.$

23. Дать общее решение задачи об охлаждении шара радиуса a при условии, что начальное распределение температуры задано

$$u(r, t)|_{t=0} = f(r),$$

а температура поверхности равна нулю. Рассмотреть случай $f(r) = T_0 = \text{const}$.

Ответ. $u(r, t) = \frac{2}{ar} \sum_{n=1}^{\infty} A_n e^{-\left(\frac{n \pi v}{a}\right)^2 t} \sin\left(\frac{n \pi r}{a}\right), \quad A_n = \int_0^a r f(r) \sin\left(\frac{n \pi r}{a}\right) dr.$

24. Найти распределение температуры в шаре радиуса a , поверхность которого, начиная с момента времени $t = 0$, излучает тепло по закону Ньютона, а начальная температура равна $T_0 = \text{const}$.

$$\text{Ответ. } u(r, t) = \frac{2aT_0}{r} \frac{ha}{1-ha} \sum_{n=1}^{\infty} \frac{\cos \gamma_n}{\gamma_n} \frac{\sin\left(\frac{\gamma_n r}{a}\right)}{1 - \frac{\sin(2\gamma_n)}{2\gamma_n}} e^{-\left(\frac{v\gamma_n}{a}\right)^2 t}, \text{ где } \gamma_n, n = 1, 2, \dots \text{ —}$$

последовательные положительные корни уравнения $\operatorname{tg}(\gamma) = \frac{\gamma}{1-ha}$; h — коэффициент теплообмена.

25. Пластина толщины $2a$ ($-a \leq x \leq a$) с заданным начальным распределением температуры $u(x, 0) = f(x)$, начиная с момента времени $t = 0$, излучает тепло в окружающую среду, температура которой принимается равной нулю. Считается, что излучение подчиняется закону Ньютона. Найти распределение температуры в пластине в произвольный момент времени.

$$\text{Ответ. } u(r, t) = \sum_{n=1}^{\infty} A_n \frac{\cos\left(\frac{\gamma_n^{(1)} x}{a}\right)}{1 + \frac{\sin(2\gamma_n^{(1)})}{2\gamma_n^{(1)}}} e^{-\left(\frac{v\gamma_n^{(1)}}{a}\right)^2 t} + \sum_{n=1}^{\infty} B_n \frac{\sin\left(\frac{\gamma_n^{(2)} x}{a}\right)}{1 - \frac{\sin(2\gamma_n^{(2)})}{2\gamma_n^{(2)}}} e^{-\left(\frac{v\gamma_n^{(2)}}{a}\right)^2 t}, \text{ где}$$

$$A_n = \frac{1}{a} \int_{-a}^a f(\xi) \cos\left(\frac{\gamma_n^{(1)} \xi}{a}\right) d\xi, \quad B_n = \frac{1}{a} \int_{-a}^a f(\xi) \sin\left(\frac{\gamma_n^{(2)} \xi}{a}\right) d\xi; \quad \gamma_n^{(1)} \text{ —}$$

последовательные положительные корни уравнения $\operatorname{tg}(\gamma^{(1)}) = \frac{ah}{\gamma^{(1)}}$; $\gamma_n^{(2)}$ —

последовательные положительные корни уравнения $\operatorname{tg}(\gamma^{(2)}) = -\frac{\gamma^{(2)}}{ah}$ h — коэффициент теплообмена.

26. Решить уравнение колебаний круглой мембраны

$$\frac{1}{v^2} \frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial r^2} + \frac{1}{r} \frac{\partial u}{\partial r} + \frac{1}{r^2} \frac{\partial^2 u}{\partial \varphi^2}$$

при краевом условии $u|_{r=b} = 0$ (мембрана закреплена по контуру) и начальных

условиях $u|_{t=0} = 0, \left. \frac{\partial u}{\partial t} \right|_{t=0} = 1 - \frac{r^2}{b^2}$.

$$\text{Ответ. } u(x, t) = \frac{8b}{v} \sum_{n=1}^{\infty} \frac{J_0\left(\frac{\alpha_n r}{b}\right)}{\alpha_n^4 J_1(\alpha_n)} \sin\left(\frac{\alpha_n v t}{b}\right), \alpha_n \text{ — } n\text{-ый положительный корень}$$

уравнения $J_0(x) = 0$.

27. Однородная струна длиной l натянута между точками $x = 0$ и $x = l$. В точке $x = c$ струна оттягивается на небольшое расстояние h от положения равновесия и в момент времени $t = 0$ отпускается без начальной скорости. Определить отклонение $u(x, t)$ струны для любого момента времени.

$$\text{Ответ. } u(x, t) = \frac{2h}{\pi^2} \frac{l^2}{c(l-c)} \sum_{k=1}^{\infty} \frac{1}{k^2} \sin\left(\frac{k\pi c}{l}\right) \sin\left(\frac{k\pi x}{l}\right) \cos\left(\frac{k\pi vt}{l}\right).$$

28. Решить задачу:

$$\frac{1}{a^2} \frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2}, \quad 0 < x < l, \quad t > 0;$$

$$u(0, t) = 0, \quad \left. \frac{\partial u}{\partial x} \right|_{x=l} = 0, \quad u(x, 0) = T_0.$$

$$\text{Ответ. } u(x, t) = \frac{4T_0}{\pi} \sum_{k=1}^{\infty} \frac{1}{2k-1} \exp\left[-v^2 \left(\frac{(2k-1)\pi}{2l}\right)^2 t\right] \sin\left(\frac{(2k-1)\pi x}{2l}\right).$$

29. Решить задачу:

$$\frac{1}{a^2} \frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}, \quad 0 < x < l, \quad t > 0;$$

$$\left. \frac{\partial u}{\partial x} \right|_{x=0} = 0, \quad \left. \frac{\partial u}{\partial x} \right|_{x=l} = 0, \quad u(x, 0) = x(l-x), \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = 0.$$

$$\text{Ответ. } u(x, t) = l^2 \left[\frac{1}{3} - \frac{1}{\pi^2} \sum_{k=1}^{\infty} \frac{1}{k^2} \cos \frac{2k\pi x}{l} \cos \frac{2k\pi vt}{l} \right].$$

30. Решить задачу Дирихле для уравнения Лапласа в квадрате $D\{0 \leq x \leq l; 0 \leq y \leq l\}$ при краевых условиях: $u(x, 0) = u(x, l) = x(l-x)$, $u(0, y) = u(l, y) = 0$.

Ответ.

$$u(x, t) = \frac{8l^2}{\pi^3} \sum_{k=1}^{\infty} \frac{1}{(2k-1)^3} \left[\operatorname{ch}\left(\frac{(2k-1)\pi y}{l}\right) + b_k \operatorname{sh}\left(\frac{(2k-1)\pi y}{l}\right) \right] \sin\left(\frac{(2k-1)\pi x}{l}\right),$$

$$\text{где } b_k = \frac{1 - \operatorname{ch}\left(\frac{(2k-1)\pi y}{l}\right)}{\operatorname{sh}\left(\frac{(2k-1)\pi y}{l}\right)}.$$

31. Решить смешанную задачу

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial}{\partial x} \left(x \frac{\partial u}{\partial x} \right), \quad 0 < x < 1,$$

$$|u|_{x=0} < \infty, \quad \left. \frac{\partial u}{\partial x} \right|_{x=1} = 0, \quad u|_{t=0} = 0, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = J_0(\mu_k \sqrt{x}),$$

где μ_k — положительный корень уравнения $J_1(\mu) = 0$.

Ответ. $u(x, t) = \frac{2}{\mu_k} \sin\left(\frac{\mu_k t}{2}\right) J_0(\mu_k \sqrt{x})$.

32. Решить смешанную задачу

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial}{\partial x} \left(x \frac{\partial u}{\partial x} \right) - \frac{9u}{4x}, \quad 0 < x < 1,$$

$$|u|_{x=0} < \infty, \quad u|_{x=1} = 0, \quad u|_{t=0} = 0, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = J_3(\mu_1 \sqrt{x}),$$

где μ_1 — положительный корень уравнения $J_3(\mu) = 0$.

Ответ. $u(x, t) = \frac{2}{\mu_1} \sin\left(\frac{\mu_1 t}{2}\right) J_3(\mu_1 \sqrt{x})$.

33. Дан неограниченный круговой цилиндр радиуса R . Найти распределение температуры внутри цилиндра в момент времени t , если на поверхности цилиндра поддерживается все время нулевая температура, а температура внутри цилиндра в начальный момент времени равна $u|_{t=0} = A J_0\left(\frac{\mu_k r}{R}\right)$, где μ_k — положительный корень уравнения $J_0(\mu) = 0$.

Ответ. $u(r, t) = A e^{-\left(\frac{v\mu_k}{R}\right)^2 t} J_0\left(\frac{\mu_k r}{R}\right)$.

34. Решить задачу 33, если с поверхности цилиндра происходит лучеиспускание в окружающую среду, температура которой равна нулю, а начальная температура равна $u|_{t=0} = u_0(r)$.

Ответ. $u(r, t) = \frac{2}{R^2} \sum_{n=1}^{\infty} \frac{a_n \mu_n^2}{\mu_n^2 + h^2 R^2} e^{-\left(\frac{v\mu_n}{R}\right)^2 t} J_0\left(\frac{\mu_n r}{R}\right)$,

$$a_n = \frac{1}{J_0^2(\mu_n)} \int_0^R r u_0(r) J_0\left(\frac{\mu_n r}{R}\right) dr, \quad \text{где } \mu_n \text{ — положительные корни}$$

уравнения $\mu J_0'(\mu) + hR J_0(\mu) = 0$.

Указание. Граничные условия имеют вид $|u|_{r=0} < \infty, \quad \left. \frac{\partial u}{\partial r} + hu \right|_{r=R} = 0$.

35. Решить смешанную задачу

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial}{\partial x} \left(x \frac{\partial u}{\partial x} \right), \quad 0 < x < \frac{1}{4},$$

$$|u|_{x=0} < \infty, \quad u|_{x=\frac{1}{4}} = 0, \quad u|_{t=0} = J_0(2\mu_1\sqrt{x}), \quad \frac{\partial u}{\partial t}\Big|_{t=0} = 0,$$

где μ_1 — положительный корень уравнения $J_0(\mu) = 0$.

Ответ. $u(x, t) = \cos(\mu_1 t) J_0(2\mu_1\sqrt{x})$.

36. Решить смешанную задачу

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + \frac{1}{x} \frac{\partial u}{\partial x} - \frac{9}{x^2} u, \quad 0 < x < 1,$$

$$|u|_{x=0} < \infty, \quad u|_{x=1} = 0, \quad u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}\Big|_{t=0} = J_3(\mu_1 x),$$

где μ_1 — положительный корень уравнения $J_3(\mu) = 0$.

Ответ. $u(x, t) = \frac{1}{\mu_1} \sin(\mu_1 t) J_3(\mu_1 x)$.

37. Решить смешанную задачу

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + \frac{1}{x} \frac{\partial u}{\partial x} - \frac{4}{x^2} u, \quad 0 < x < 1,$$

$$|u|_{x=0} < \infty, \quad u|_{x=1} = 0, \quad u|_{t=0} = J_2(\mu_k x), \quad \frac{\partial u}{\partial t}\Big|_{t=0} = J_2(\mu_k x),$$

где μ_k — положительный корень уравнения $J_2(\mu) = 0$.

Ответ. $u(x, t) = \left[\frac{1}{\mu_k} \sin(\mu_k t) + \cos(\mu_k t) \right] J_2(\mu_k x)$.

38. Решить смешанную задачу

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + \frac{1}{x} \frac{\partial u}{\partial x} - \frac{4}{x^2} u, \quad 0 < x < 1,$$

$$|u|_{x=0} < \infty, \quad u|_{x=1} = 0, \quad u|_{t=0} = \frac{1}{2} J_2(\mu_k x), \quad \frac{\partial u}{\partial t}\Big|_{t=0} = \frac{3}{2} J_2(\mu_k x),$$

где μ_k — положительный корень уравнения $J_2(\mu) = 0$.

Ответ. $u(x, t) = \left[\frac{3}{2\mu_k} \sin(\mu_k t) + \frac{1}{2} \cos(\mu_k t) \right] J_2(\mu_k x).$

39. Решить смешанную задачу

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + \frac{1}{x} \frac{\partial u}{\partial x} - \frac{u}{x^2}, \quad 0 < x < 1,$$

$$\left| u \right|_{x=0} < \infty, \quad u|_{x=1} = 0, \quad u|_{t=0} = J_1(\mu_k x) + J_1(\mu_m x), \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = 0,$$

где μ_k и μ_m — два различных положительных корня уравнения $J_1(\mu) = 0$.

Ответ. $u(x, t) = J_1(\mu_k x) \cos(\mu_k t) + J_1(\mu_m x) \cos(\mu_m t).$

40. Решить смешанную задачу

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + \frac{1}{x} \frac{\partial u}{\partial x} - \frac{u}{x^2}, \quad 0 < x < 1,$$

$$\left| u \right|_{x=0} < \infty, \quad u|_{x=1} = 0, \quad u|_{t=0} = J_1(\mu_k x), \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = J_1(\mu_m x),$$

где μ_k и μ_m — два различных положительных корня уравнения $J_1(\mu) = 0$.

Ответ. $u(x, t) = J_1(\mu_k x) \cos(\mu_k t) + \frac{1}{\mu_m} J_1(\mu_m x) \sin(\mu_m t).$

5. Неоднородные задачи математической физики

В предыдущей главе мы рассматривали *однородные задачи* математической физики с разделяющимися переменными: основное разрешающее уравнение задачи и обязательно дополнительные условия (начальные или граничные условия) по одной из переменных были однородными.

Содержание данной главы составляют *неоднородные задачи* математической физики с разделяющимися переменными, приводящиеся к интегрированию уравнения

$$L_x(u) + M_y(u) = F(x, y), \quad a < x < b, \quad c < y < d, \quad (5.1)$$

где L_x и M_y — линейные дифференциальные операторы

$$L_x(u) = \frac{1}{r(x)} \left[\frac{\partial}{\partial x} \left(p(x) \frac{\partial u}{\partial x} \right) - q(x)u \right],$$

$$M_y(u) = A(y) \frac{\partial^2 u}{\partial y^2} + B(y) \frac{\partial u}{\partial y} + C(y)u,$$

при неоднородных граничных условиях первого, второго или третьего рода по переменной x

$$u|_{x=a} = f_a(y), \quad u|_{x=b} = f_b(y), \quad (5.2-I)$$

$$\frac{\partial u}{\partial x} \Big|_{x=a} = f_a(y), \quad \frac{\partial u}{\partial x} \Big|_{x=b} = f_b(y), \quad (5.2-II)$$

$$\frac{\partial u}{\partial x} - h_a u \Big|_{x=a} = f_a(y), \quad \frac{\partial u}{\partial x} + h_b u \Big|_{x=b} = f_b(y) \quad (5.2-III)$$

и условиях по переменной y , в зависимости от типа уравнения (5.1). Например, если $A > 0$, то уравнение (5.1) эллиптического типа; в этом случае ставится одно из условий первого, второго или третьего рода, то есть

$$u|_{y=c} = \varphi_c(x), \quad u|_{y=d} = \varphi_d(x), \quad (5.3-I)$$

$$\frac{\partial u}{\partial y} \Big|_{y=c} = \varphi_c(x), \quad \frac{\partial u}{\partial y} \Big|_{y=d} = \varphi_d(x), \quad (5.3-II)$$

$$\frac{\partial u}{\partial y} - h_c u \Big|_{y=c} = \varphi_c(x), \quad \frac{\partial u}{\partial y} + h_d u \Big|_{y=d} = \varphi_d(x). \quad (5.3-III)$$

Если $A < 0$, то тип уравнения (5.1) гиперболический; тогда $y \in (c, +\infty)$ и ставятся условия

$$u|_{y=c} = \varphi(x), \quad \left. \frac{\partial u}{\partial y} \right|_{y=c} = \psi(x); \quad (5.4)$$

если $A = 0$, то тип уравнения (5.1) параболический; $y \in (c, +\infty)$, и ставится условие

$$u|_{y=c} = \varphi(x). \quad (5.5)$$

Здесь все коэффициенты и правые части уравнений — заданные функции; относительно коэффициентов уравнений сохраняются предположения главы 4.

Неоднородные задачи характерны тем, что они описываются неоднородным уравнением и неоднородными дополнительными условиями — граничными или начальными условиями. Это означает, что правые части уравнений — не нули, а заданные функции. Левые части уравнений сохраняют структуру однородной задачи. Ясно, что к таким задачам метод Фурье разделения переменных неприменим.

Для решения неоднородных задач служат два метода:

- метод приведения к однородной задаче;
- метод Гринберга или метод конечных интегральных преобразований.

Метод приведения к однородной задаче

В тех случаях, когда можно подобрать частное решение u_1 уравнения (5.1), удовлетворяющее условиям (5.2), рассматриваемая задача при помощи подстановки $u = u_1 + u_2$, приводится к однородной, и для решения ее может быть использован метод Фурье. Такой подход требует применения искусственных приемов. Его целесообразно использовать в простых случаях, когда легко бывает выделить частное решение.

Задача просто решается также и в том случае, когда неоднородным является только дифференциальное уравнение, а граничные условия однородны ($f_a = f_b = 0$). Решение задачи в рассматриваемом случае ищется в форме разложения по собственным функциям соответствующей однородной задачи

$$u = \sum_{n=1}^{\infty} u_n(y) X_n(x), \quad a < x < b, \quad (5.6)$$

где $X_n(x)$ — нетривиальные решения дифференциального уравнения

$$(pX')' + (\lambda r - q)X = 0, \quad (5.7)$$

удовлетворяющие однородным граничным условиям одного из типов

$$X|_{x=a} = 0, \quad X|_{x=b} = 0, \quad (5.8-I)$$

$$\left. \frac{dX}{dx} \right|_{x=a} = 0, \quad \left. \frac{dX}{dx} \right|_{x=b} = 0, \quad (5.8-II)$$

$$\left. \frac{dX}{dx} - h_a X \right|_{x=a} = 0, \quad \left. \frac{dX}{dx} + h_b X \right|_{x=b} = 0. \quad (5.8-III)$$

В предположении, что правая часть уравнения допускает разложение в ряд по функциям $X_n(x)$

$$F = \sum_{n=1}^{\infty} F_n(y) X_n(x), \quad a < x < b,$$

$$F_n(y) = \frac{\int_a^b rF(x,y) X_n(x) dx}{\int_a^b rX_n^2(x) dx},$$

определение коэффициентов $u_n(y)$ приводится, после подстановки ряда (5.6) в уравнение (5.1), к интегрированию обыкновенного дифференциального уравнения

$$M_y(u_n) - \lambda_n u_n = F_n(y),$$

где λ_n — собственные значения однородной задачи.

Постоянные интегрирования находятся при этом из соотношений, которые получаются, если подставить (5.6) в (5.3) (или, соответственно, в (5.4), (5.5)) и, разложив входящие в правые части этих равенств функции по собственным функциям, приравнять затем коэффициенты при $X_n(x)$.

Общий случай, когда граничные условия неоднородны, может быть приведен к рассмотренному случаю задачи с неоднородным уравнением и однородными граничными условиями, если искать решение в виде суммы $u = u_1 + u_2$, где u_1 — какая-либо достаточно гладкая функция, удовлетворяющая условиям (5.2), но, в отличие от предыдущего, не обязательно являющаяся решением уравнения.

Например, при граничных условиях первого рода

$$u|_{x=a} = f_a(y), \quad u|_{x=b} = f_b(y),$$

в качестве такой функции можно выбрать линейную относительно x функцию

$$u_1 = \frac{x-a}{b-a} f_b(y) + \frac{b-x}{b-a} f_a(y),$$

при условиях второго рода — функцию

$$u_1 = \frac{1}{2} \frac{(x-a)^2}{b-a} f_b(y) - \frac{1}{2} \frac{(b-x)^2}{b-a} f_a(y),$$

и т.д.¹

Следует заметить, что подобный метод приведения, связанный с введением в значительной мере произвольной функции u_1 , не всегда является удобным. Например, в тех случаях, когда граничные функции разрывны и при неудачном выборе этой функции такой метод может повести к значительному усложнению последующих этапов решения.

Метод Гринберга

Более адекватный метод решения неоднородных задач, свободный от необходимости прибегать в каждом случае к подбору функции u_1 , требующему в некоторых случаях значительной изобретательности, предложен Г.А. Гринбергом² и известен как метод Гринберга или метод конечных интегральных преобразований. Можно считать, что этот метод является обобщением метода Фурье на случай неоднородного уравнения и неоднородных граничных условий. В тех случаях, когда граничные условия однородны и неоднородным является лишь дифференциальное уравнение, метод Гринберга приводит к тем же формулам, что и классический метод решения задач этого типа.

Идея метода Гринберга заключается в том, что решение неоднородной задачи так же, как и выше, ищется в виде ряда (5.6), коэффициенты которого, в соответствии с общей теорией разложения по собственным функциям, могут быть представлены формулами

$$u_n(y) = \frac{\int_a^b r(x)u(x,y)X_n(x)dx}{\int_a^b r(x)X_n^2(x)dx} = \frac{\bar{u}_n(y)}{\|X_n\|^2}.$$

Для получения формального решения задачи достаточно найти значение интеграла \bar{u}_n (эта функция иногда называется *трансформантой* от функции $u(x,y)$). Это может быть сделано с помощью следующего приема. Умножим уравнение (5.1) на $r(x)X_n(x)$ и проинтегрируем по интервалу $[a,b]$. Получим

$$\int_a^b X_n \left[\frac{\partial}{\partial x} \left(p \frac{\partial u}{\partial x} \right) - qu \right] dx + \int_a^b M_y(u) r(x) X_n(x) dx =$$

¹ Важно отметить, что выбор функции u_1 в указанной форме в ряде случаев может привести к значительному усложнению хода решения задачи.

² Гринберг Г.А. Избранные вопросы математической теории электрических и магнитных явлений. Изд АН СССР, 1948.

$$= \int_a^b F(x, y) r(x) X_n(x) dx. \quad (5.9)$$

Первый интеграл слева преобразуем по частям

$$\begin{aligned} \int_a^b X_n \left[\frac{\partial}{\partial x} \left(p \frac{\partial u}{\partial x} \right) \right] dx &= X_n p \frac{\partial u}{\partial x} \Big|_{x=a}^{x=b} - \int_a^b p \frac{\partial u}{\partial x} X_n' dx = \\ &= X_n p \frac{\partial u}{\partial x} \Big|_{x=a}^{x=b} - p X_n' u \Big|_{x=a}^{x=b} + \int_a^b u (p X_n')' dx. \end{aligned}$$

Теперь, вместо формулы (5.9), будем иметь:

$$\begin{aligned} p \left[X_n \frac{\partial u}{\partial x} - X_n' u \right] \Big|_{x=a}^{x=b} + \int_a^b \underbrace{(p X_n')' - q X_n}_{=-\lambda_n r X_n} dx + \\ + M_y \left(\int_a^b r X_n u dx \right) = \underbrace{\int_a^b F(x, y) r X_n dx}_{=\bar{F}_n(y)}. \end{aligned}$$

С учетом уравнения (5.7)

$$(p X_n')' - q X_n = -\lambda_n r X_n$$

и обозначения

$$\int_a^b F(x, y) r(x) X_n(x) dx = \bar{F}_n(y)$$

получаем уравнение для трансформанты $\bar{u}_n(y)$

$$M_y(\bar{u}_n) - \lambda_n \bar{u}_n = \bar{F}_n(y) - p \left[X_n \frac{\partial u}{\partial x} - X_n' u \right] \Big|_{x=a}^{x=b}. \quad (5.10)$$

Уравнение (5.10) — обыкновенное дифференциальное уравнение относительно трансформанты $\bar{u}_n(y)$. Правая часть этого уравнения содержит только известные функции. Поэтому уравнение (5.10) может служить для определения $\bar{u}_n(y)$. Постоянные интегрирования находятся из уравнений, получающихся из (5.3) (или (5.4), (5.5)) путем того же приема — умножения на $r(x) X_n(x)$ и интегрирования по промежутку $[a, b]$.

Всё сказанное выше переносится на случай условий четвертого рода, а также на случай сингулярной задачи Штурма-Лиувилля, если её спектр дискретный. Метод допускает распространение на некоторые задачи математической физики, связанные

с уравнениями высших порядков, или на задачи, в которых искомая функция зависит от большего числа переменных.

Следует отметить, что при неоднородных условиях первого рода ряды, представляющие решение, сходятся неравномерно вблизи концов интервала.³ В случае условий второго рода аналогичное явление имеет место для ряда, представляющего производную $\frac{\partial u}{\partial x}$.

Краткая схема решения задачи методом Гринберга

Приведем краткую схему, которой следует придерживаться при решении задач математической физики методом Гринберга.

1. Поставить задачу Штурма-Лиувилля, связанную с рассматриваемой задачей и найти систему собственных значений и собственных функций.
2. Трансформировать исходное уравнение в частных производных в обыкновенное дифференциальное уравнение для трансформанты $\bar{u}_n(y)$.
3. Преобразовать граничные условия по второй переменной в условия для трансформанты $\bar{u}_n(y)$.
4. Найти трансформанту $\bar{u}_n(y)$ и получить решение исходной задачи в виде разложения по собственным функциям задачи Штурма-Лиувилля.
5. Рассмотреть вопрос о сходимости полученного формального ряда и, в случае необходимости, улучшить её.

Примеры решения типовых задач

Задача 5.1. Решить смешанную задачу для неоднородного уравнения гиперболического типа

$$\frac{\partial^2 u}{\partial t^2} = v^2 \frac{\partial^2 u}{\partial x^2} + g, \quad v^2 = \frac{E}{\rho};$$

$$u|_{t=0} = 0, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = 0; \quad u|_{x=0} = 0, \quad \left. \frac{\partial u}{\partial x} \right|_{x=l} = 0.$$

Физическая интерпретация задачи: стержень подвешен вертикально и закреплен так, что смещение во всех сечениях равно нулю. В момент времени $t = 0$ стержень освобождается, оставаясь закрепленным на верхнем конце. Требуется изучить вынужденные колебания стержня под действием силы тяжести.

Решение. Решаем задачу в Марле. Определяем уравнение и дополнительные условия

³ Если условие неоднородно только на конце $x = a$, то, соответственно, только на этом конце.

> eq:=diff(u(x,t),t\$2)=v^2*diff(u(x,t),x\$2)+g; 0<x,x<a,t>0;

$$eq := \frac{\partial^2}{\partial t^2} u(x, t) = v^2 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) + g$$

$$0 < x, x < a, 0 < t$$

> init_c:= [u(x,0)=0,D[2](u)(x,0)=0];

$$init_c := [u(x, 0) = 0, D_2(u)(x, 0) = 0]$$

> bound_c:= [u(0,t)=0,D[1](u)(l,t)=0];

$$bound_c := [u(0, t) = 0, D_1(u)(l, t) = 0]$$

Задачу будем решать методом Гринберга. Сформулируем соответствующую задачу Штурма-Лиувилля. Для этого произведем разделение переменных в однородном уравнении (при $g = 0$)

> pdsolve(subs(g=0,eq),HINT=X(x)*T(t));

$$(u(x, t) = X(x) T(t)) \&where \left[\left\{ \frac{d^2}{dx^2} X(x) = -c_1 X(x), \frac{d^2}{dt^2} T(t) = v^2 -c_1 T(t) \right\} \right]$$

Соответствующая задача Штурма-Лиувилля будет такой:

> odel:=subs(_c[1]=-lambda,op([2,1,1],%));

$$bound := [X(0) = 0, D(X)(l) = 0];$$

$$odel := \frac{d^2}{dx^2} X(x) = -\lambda X(x)$$

$$bound := [X(0) = 0, D(X)(l) = 0]$$

Находим общее решение уравнения Штурма-Лиувилля

> sol_ode:=dsolve(odel,X(x));

$$sol_ode := X(x) = -C1 \sin(\sqrt{\lambda} x) + -C2 \cos(\sqrt{\lambda} x)$$

> X:=unapply(rhs(sol_ode),x);

$$X := x \rightarrow -C1 \sin(\sqrt{\lambda} x) + -C2 \cos(\sqrt{\lambda} x)$$

Подставим это решение в однородные граничные условия и получим уравнение для определения собственных значений

> sist:={bound[1],bound[2]};

$$sist := \{ -C2 = 0, -C1 \cos(\sqrt{\lambda} l) \sqrt{\lambda} - C2 \sin(\sqrt{\lambda} l) \sqrt{\lambda} = 0 \}$$

> A:=linalg[genmatrix](sist,[_C1,_C2]);

$$A := \begin{bmatrix} 0 & 1 \\ \cos(\sqrt{\lambda} l) \sqrt{\lambda} & -\sin(\sqrt{\lambda} l) \sqrt{\lambda} \end{bmatrix}$$

> Delta:=linalg[det](A);

$$\Delta := -\cos(\sqrt{\lambda} l) \sqrt{\lambda}$$

Здесь мы сформировали матрицу системы уравнений по граничным условиям и вычислили определитель этой матрицы.

Таким образом, приравняв нулю найденный определитель, получаем характеристическое уравнение для определения собственных значений

$$\cos(\sqrt{\lambda} l) = 0.$$

Читателю мы предоставляем возможность самостоятельно убедиться, что $\lambda = 0$ не является собственным значением нашей задачи.

Решаем полученное характеристическое уравнение и находим спектр задачи

```
> _EnvAllSolutions:=true:
solve(simplify(Delta/sqrt(lambda))=0, lambda);
```

$$\frac{\pi^2 (1 + 2 \sqrt{Zl})^2}{4 l^2}$$

Таким образом, собственные значения будут

$$\lambda = \lambda_k = \left[\frac{(2k+1)\pi}{2l} \right]^2, \quad k = 0, 1, 2, 3, \dots$$

Определяем их в Maple

```
> ev:=unapply(subs(_Z1=k,%),k);
```

$$ev := k \rightarrow \frac{1}{4} \frac{\pi^2 (1 + 2 k)^2}{l^2}$$

Найдем теперь собственные функции задачи

```
> X:='X':subs(lambda=ev(k),ode1);
assume(k, posint):
dsolve({%,bound[1],bound[2]},X(x));
```

$$\frac{d^2}{dx^2} X(x) = \frac{1}{4} \frac{\pi^2 (1 + 2 k)^2 X(x)}{l^2}$$

$$X(x) = _C1 \sin\left(\frac{\pi (1 + 2 k) x}{2 l}\right)$$

Сразу же и нормируем их

```
> rhs(%)/_C1;
X:=unapply(%/simplify(int(%^2,x=0..1)^(1/2)),k,x);
```

$$\sin\left(\frac{\pi (1 + 2 k) x}{2 l}\right)$$

$$X := (k, x) \rightarrow \frac{\sin\left(\frac{1}{2} \frac{\pi (1 + 2k) x}{l}\right) \sqrt{2}}{\sqrt{l}}$$

Итак, нормированные собственные функции имеют вид

$$X_k(x) = \sqrt{\frac{2}{l}} \sin(\sqrt{\lambda_k} x) = \sqrt{\frac{2}{l}} \sin\left(\frac{(2k+1)\pi x}{2l}\right).$$

В соответствии с методом Гринберга решение задачи ищем в виде ряда

$$u(x, t) = \sum_k u_k(t) X_k(x).$$

Для определения трансформанты $u_k(t)$ умножим исходное уравнение задачи на собственную функцию $X_k(x)$ и проинтегрируем по отрезку $[0, l]$. Выполним все необходимые преобразования в Maple

```
> ode2 := Int (lhs (eq) * X (lambda, x), x=0..l) =
      Int (op (1, rhs (eq)) * X (lambda, x), x=0..l) +
      int (op (2, rhs (eq)) * X (lambda, x), x=0..l);
```

$$\begin{aligned} \text{ode2} &:= \int_0^l \left(\frac{\partial^2}{\partial t^2} u(x, t) \right) \sqrt{2} \sqrt{\frac{1}{l}} \sin(\sqrt{\lambda} x) dx = \\ &\int_0^l v^2 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) \sqrt{2} \sqrt{\frac{1}{l}} \sin(\sqrt{\lambda} x) dx - \frac{g \sqrt{2} \sqrt{\frac{1}{l}} (\cos(\sqrt{\lambda} l) - 1)}{\sqrt{\lambda}} \end{aligned}$$

Обозначим

$$u_k(t) = \int_0^l u(x, t) X_k(x) dx.$$

Тогда (временно обозначим $u_k(t) = U(t)$)

```
> ode2 :=
      subs ({lhs (ode2) = diff (U (t), t$2), cos (sqrt (lambda) * l) = 0},
      ode2);
```

$$\text{ode2} := \frac{d^2}{dt^2} U(t) = \int_0^l v^2 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) \sqrt{2} \sqrt{\frac{1}{l}} \sin(\sqrt{\lambda} x) dx + \frac{g \sqrt{2} \sqrt{\frac{1}{l}}}{\sqrt{\lambda}}$$

Интеграл в правой части берем два раза по частям. Для этого сначала выделяем этот интеграл в отдельное выражение

> op(1, rhs(ode2));

$$\int_0^l v^2 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) \sqrt{2} \sqrt{\frac{1}{l}} \sin(\sqrt{\lambda} x) dx$$

Теперь преобразовываем его два раза по частям с помощью команды `intparts` пакета `student`. После первого преобразования получим

> student[intparts](%, sin(sqrt(lambda)*x));

$$\sin(\sqrt{\lambda} l) v^2 \sqrt{2} \sqrt{\frac{1}{l}} D_1(u)(l, t) - \int_0^l \cos(\sqrt{\lambda} x) \sqrt{\lambda} v^2 \sqrt{2} \sqrt{\frac{1}{l}} \left(\frac{\partial}{\partial x} u(x, t) \right) dx$$

Наконец, после второго преобразования, получим

> student[intparts](%, cos(sqrt(lambda)*x));

$$\begin{aligned} & \sin(\sqrt{\lambda} l) v^2 \sqrt{2} \sqrt{\frac{1}{l}} D_1(u)(l, t) - \cos(\sqrt{\lambda} l) \sqrt{\lambda} v^2 \sqrt{2} \sqrt{\frac{1}{l}} u(l, t) \\ & + \sqrt{\lambda} v^2 \sqrt{2} \sqrt{\frac{1}{l}} u(0, t) + \int_0^l -\sin(\sqrt{\lambda} x) \lambda v^2 \sqrt{2} \sqrt{\frac{1}{l}} u(x, t) dx \end{aligned}$$

Учтем теперь в полученном выражении граничные условия

> subs({cos(sqrt(lambda)*1)=0, bound_c[1], bound_c[2]}, %);

$$\int_0^l -\sin(\sqrt{\lambda} x) \lambda v^2 \sqrt{2} \sqrt{\frac{1}{l}} u(x, t) dx$$

Таким образом, мы получили

> ode2:=subs(op(1, rhs(ode2))=%, ode2);

$$ode2 := \frac{d^2}{dt^2} U(t) = \int_0^l -\sin(\sqrt{\lambda} x) \lambda v^2 \sqrt{2} \sqrt{\frac{1}{l}} u(x, t) dx + \frac{g \sqrt{2} \sqrt{\frac{1}{l}}}{\sqrt{\lambda}}$$

Или, с учетом обозначения для трансформанты, окончательно получаем следующее обыкновенное дифференциальное уравнение для определения $u_k(t)$

> ode2:=subs(op(1, rhs(ode2))=-v^2*lambda*U(t), ode2);

$$ode2 := \frac{d^2}{dt^2} U(t) = -v^2 \lambda U(t) + \frac{g \sqrt{2} \sqrt{\frac{1}{l}}}{\sqrt{\lambda}}$$

Решаем теперь это уравнение с учетом начальных условий

> sol_ode2:=dsolve({ode2,U(0)=0,D(U)(0)=0},U(t));

$$sol_ode2 := U(t) = -\frac{\cos(v\sqrt{\lambda}t)\sqrt{2}\sqrt{\frac{1}{l}}g}{\lambda^{(3/2)}v^2} + \frac{\sqrt{2}\sqrt{\frac{1}{l}}g}{\lambda^{(3/2)}v^2}$$

Итак, мы получили

> assume(l>0):

U:=unapply(simplify(subs(lambda=ev(k),rhs(sol_ode2))),k,t);

$$U := (k, t) \rightarrow -\frac{8\sqrt{2}l^{(5/2)}g\left(\cos\left(\frac{1}{2}\frac{v\pi(1+2k)t}{l}\right)-1\right)}{\pi^3(1+2k)^3v^2}$$

или, возвращаясь к старым обозначениям

$$u_k(t) = \frac{8\sqrt{2}gl^{5/2}\left[1 - \cos\left(\frac{(2k+1)\pi vt}{2l}\right)\right]}{v^2\pi^3(2k+1)^3}.$$

Таким образом, решение задачи имеет вид

> 'u(x,t)'=Sum(U(k,t)*X(ev(k),x),k=0..infinity);

$$u(x,t) = \sum_{k=0}^{\infty} \left(-\frac{16l^{-2}g\left(\cos\left(\frac{v\pi(1+2k)t}{2l}\right)-1\right)\sin\left(\frac{\pi(1+2k)x}{2l}\right)}{\pi^3(1+2k)^3v^2} \right)$$

Окончательно, будем иметь

$$u(x,t) = \frac{16gl^2}{\pi^3v^2} \sum_{k=0}^{\infty} \frac{\left[1 - \cos\left(\frac{(2k+1)\pi vt}{2l}\right)\right]}{(2k+1)^3} \sin\left(\frac{(2k+1)\pi x}{2l}\right).$$

Найденное решение можно преобразовать к другому виду, если просуммировать часть полученного выражения. А именно, покажем, что справедлива формула

$$\frac{gx(2l-x)}{2v^2} = \frac{16gl^2}{\pi^3v^2} \sum_{k=0}^{\infty} \frac{\sin\left(\frac{(2k+1)\pi x}{2l}\right)}{(2k+1)^3}.$$

Эта формула легко устанавливается разложением функции, стоящей слева, в ряд Фурье по собственным функциям нашей задачи Штурма-Лиувилля. Действительно, определим эту функцию в Maple

> f:=x->g*x*(2*1-x)/2/v^2;

$$f := x \rightarrow \frac{1}{2} \frac{gx(2l-x)}{v^2}$$

Тогда ее коэффициенты Фурье вычисляются по формуле

> Int (f (x) * X (ev (k) , x) , x=0 .. l) ;

$$\int_0^l \frac{1}{2} \frac{g x (2l - x) \sqrt{2} \sin\left(\frac{\pi (1 + 2k) x}{2l}\right)}{v^2 \sqrt{l}} dx$$

> C := unapply (factor (value (%)) , k) ;

$$C := k \rightarrow \frac{8 l^{(5/2)} g \sqrt{2}}{\pi^3 (1 + 2k)^3 v^2}$$

Таким образом, мы показали, что

> f (x) = Sum (C (k) * X (ev (k) , x) , k=0 .. infinity) ;

$$\frac{g x (2l - x)}{2 v^2} = \sum_{k=0}^{\infty} \left(\frac{16 l^{-2} g \sin\left(\frac{\pi (1 + 2k) x}{2l}\right)}{\pi^3 (1 + 2k)^3 v^2} \right)$$

Следовательно, решение задачи можно записать и в такой форме

> ' u (x , t) ' = f (x) -

16 * l ^ 2 * g * Sum (cos (1 / 2 * v * Pi * (1 + 2 * k) / l * t) / (1 + 2 * k) ^ 3 * sin (1 / 2 * Pi * (1 + 2 * k) / l * x) / Pi ^ 3 / v ^ 2 , k=0 .. infinity) ;

u (x , t) =

$$\frac{g x (2l - x)}{2 v^2} - 16 l^{-2} g \left(\sum_{k=0}^{\infty} \frac{\cos\left(\frac{v \pi (1 + 2k) t}{2l}\right) \sin\left(\frac{\pi (1 + 2k) x}{2l}\right)}{(1 + 2k)^3 \pi^3 v^2} \right)$$

Итак,

$$u(x, t) = \frac{gx(2l - x)}{2v^2} - \frac{16gl^2}{\pi^3 v^2} \sum_{k=0}^{\infty} \frac{\cos\left(\frac{(2k+1)\pi vt}{2l}\right)}{(2k+1)^3} \sin\left(\frac{(2k+1)\pi x}{2l}\right).$$

Отметим, что вторая форма решения сразу получается, если искать это решение методом сведения задачи к однородной задаче, то есть в форме

$$u(x, t) = u_0(x) + w(x, t),$$

где $u_0(x)$ — стационарное решение следующей задачи

$$v^2 \frac{d^2 u_0}{dx^2} + g = 0;$$

$$u_0|_{x=0} = 0, \quad \frac{du_0}{dx}|_{x=l} = 0.$$

Легко видеть, что

$$u_0(x) = \frac{gx(2l-x)}{2v^2}.$$

Тогда для определения функции $w(x, t)$ будем иметь следующую однородную задачу

$$\frac{\partial^2 w}{\partial t^2} = v^2 \frac{\partial^2 w}{\partial x^2};$$

$$w|_{t=0} = -u_0(x), \quad \frac{\partial w}{\partial t}|_{t=0} = 0; \quad w|_{x=0} = 0, \quad \frac{\partial w}{\partial x}|_{x=l} = 0,$$

которая легко решается методом Фурье

$$w(x, t) = -\frac{16gl^2}{\pi^3 v^2} \sum_{k=1}^{\infty} \frac{\cos\left(\frac{(2k+1)\pi vt}{2l}\right)}{(2k+1)^3} \sin\left(\frac{(2k+1)\pi x}{2l}\right).$$

Мы предоставляем читателю самостоятельно проделать все необходимые выкладки (конечно, с помощью Maple!) и убедиться в этом.

Задача 5.2. На струну длиной l постоянно действует внешняя возмущающая сила, равная силе тяжести G . Найти закон колебаний струны, если начальное отклонение и начальная скорость равны нулю, а концы струны закреплены.

Решение. Сформулируем математическую задачу: найти функцию $u(x, t)$, удовлетворяющую следующему уравнению колебаний струны

$$\frac{\partial^2 u}{\partial t^2} = v^2 \frac{\partial^2 u}{\partial x^2} - G, \quad v^2 = \frac{T_0}{\rho}$$

однородным начальным

$$u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}|_{t=0} = 0$$

и однородным граничным условиям

$$u|_{x=0} = 0, \quad u|_{x=l} = 0.$$

Как видим, задача отличается от только что рассмотренной задачи 5.1 граничными условиями. Решаем задачу методом Гринберга в Maple — задаем уравнение и дополнительные условия

> eq:=diff(u(x,t),t\$2)=v^2*diff(u(x,t),x\$2)-G; 0<x,x<a,t>0;

$$eq := \frac{\partial^2}{\partial t^2} u(x, t) = v^2 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) - G$$

$$0 < x, x < a, 0 < t$$

```
> init_c := [u(x, 0) = 0, D[2](u)(x, 0) = 0];
      init_c := [u(x, 0) = 0, D_2(u)(x, 0) = 0]
> bound_c := [u(0, t) = 0, u(l, t) = 0];
      bound_c := [u(0, t) = 0, u(l, t) = 0]
```

Производим разделение переменных в однородном уравнении и формулируем соответствующую задачу Штурма-Лиувилля

```
> pdsolve(subs(G=0, eq), HINT=X(x)*T(t));
(u(x, t) = X(x) T(t)) &where [ { d^2/dx^2 X(x) = -c_1 X(x), d^2/dt^2 T(t) = v^2 -c_1 T(t) } ]
> ode1 := subs(_c[1] = -lambda, op([2, 1, 1], %));
      bound := [X(0) = 0, X(l) = 0];
      ode1 := d^2/dx^2 X(x) = -lambda X(x)
      bound := [X(0) = 0, X(l) = 0]
```

Находим общее решение обыкновенного дифференциального уравнения

```
> sol_ode := dsolve(ode1, X(x));
      sol_ode := X(x) = _C1 sin(sqrt(lambda) x) + _C2 cos(sqrt(lambda) x)
> X := unapply(rhs(sol_ode), x);
      X := x -> _C1 sin(sqrt(lambda) x) + _C2 cos(sqrt(lambda) x)
```

Подставляем это решение в однородные граничные условия; формируем матрицу коэффициентов линейной алгебраической системы и вычисляем ее определитель

```
> sist := {bound[1], bound[2]};
      sist := {_C1 sin(sqrt(lambda) l) + _C2 cos(sqrt(lambda) l) = 0, _C2 = 0}
> A := linalg[genmatrix](sist, {_C1, _C2});
      A := [ sin(sqrt(lambda) l)  cos(sqrt(lambda) l)
            0                    1 ]
> Delta := linalg[det](A);
      Delta := sin(sqrt(lambda) l)
```

Таким образом, характеристическое уравнение для определения собственных значений имеет вид $\sin(\sqrt{\lambda}l) = 0$. Откуда находим собственные значения

```
> _EnvAllSolutions := true: solve(Delta=0, lambda);
```

$$\frac{\pi^2 Z l^2}{l^2}$$

> ev:=unapply(subs(_Z1=k,%),k);

$$ev := k \rightarrow \frac{\pi^2 k^2}{l^2}$$

В результате получили дискретный спектр собственных значений

$$\lambda = \lambda_k = \frac{\pi^2 k^2}{l^2}, k = 1, 2, 3, \dots$$

Читателю мы рекомендуем самостоятельно убедиться в том, что $\lambda = 0$ не является собственным значением нашей задачи.

Определяем теперь соответствующие собственные функции задачи

> X:='X':subs(lambda=ev(k),ode1);

assume(k, posint):dsolve({%,bound[1],bound[2]},X(x));

$$\frac{d^2}{dx^2} X(x) = -\frac{\pi^2 k^2 X(x)}{l^2}$$

$$X(x) = _C1 \sin\left(\frac{\pi k x}{l}\right)$$

Сразу же и нормируем их

> rhs(%)/_C1;

X:=unapply(%/simplify(int(%^2,x=0..l)^(1/2)),k,x);

$$\sin\left(\frac{\pi k x}{l}\right)$$

$$X := (k, x) \rightarrow \frac{\sin\left(\frac{\pi k x}{l}\right) \sqrt{2}}{\sqrt{l}}$$

Итак, нормированные собственные функции будут

$$X_k(x) = \sqrt{\frac{2}{l}} \sin\left(\frac{\pi k x}{l}\right) = \sqrt{\frac{2}{l}} \sin(\sqrt{\lambda_k} x).$$

Для удобства записи дальнейших преобразований переопределим собственные функции в Maple в соответствии со вторым равенством в последней формуле

> X:=(lambda,x)->sqrt(2/l)*sin(sqrt(lambda)*x);

$$X := (\lambda, x) \rightarrow \sqrt{\frac{2}{l}} \sin(\sqrt{\lambda} x)$$

Применяем теперь процедуру метода Гринберга к исходному уравнению — умножаем его на собственные функции и интегрируем по отрезку $[0, l]$

```
> ode2 := Int (lhs (eq) * X (lambda, x), x=0..1) =
  Int (op (1, rhs (eq)) * X (lambda, x), x=0..1) +
  int (op (2, rhs (eq)) * X (lambda, x), x=0..1) ;
```

$$ode2 := \int_0^l \left(\frac{\partial^2}{\partial t^2} u(x, t) \right) \sqrt{2} \sqrt{\frac{1}{l}} \sin(\sqrt{\lambda} x) dx =$$

$$\int_0^l v^2 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) \sqrt{2} \sqrt{\frac{1}{l}} \sin(\sqrt{\lambda} x) dx + \frac{G \sqrt{2} \sqrt{\frac{1}{l}} (\cos(\sqrt{\lambda} l) - 1)}{\sqrt{\lambda}}$$

Обозначим трансформанту Гринберга

$$u_k(t) = \int_0^l u(x, t) X_k(x) dx .$$

Тогда (здесь $u_k(t) = U(t)$)

```
> ode2 := subs (lhs (ode2) = diff (U (t), t$2), ode2) ;
```

$$ode2 := \frac{d^2}{dt^2} U(t) =$$

$$\int_0^l v^2 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) \sqrt{2} \sqrt{\frac{1}{l}} \sin(\sqrt{\lambda} x) dx + \frac{G \sqrt{2} \sqrt{\frac{1}{l}} (\cos(\sqrt{\lambda} l) - 1)}{\sqrt{\lambda}}$$

Интеграл справа преобразовываем два раза по частям

```
> op (1, rhs (ode2)) ;
```

$$\int_0^l v^2 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) \sqrt{2} \sqrt{\frac{1}{l}} \sin(\sqrt{\lambda} x) dx$$

```
> student [intparts] (% , sin (sqrt (lambda) * x)) ;
```

$$\sin(\sqrt{\lambda} l) v^2 \sqrt{2} \sqrt{\frac{1}{l}} D_1(u)(l, t) - \int_0^l \cos(\sqrt{\lambda} x) \sqrt{\lambda} v^2 \sqrt{2} \sqrt{\frac{1}{l}} \left(\frac{\partial}{\partial x} u(x, t) \right) dx$$

```
> student [intparts] (% , cos (sqrt (lambda) * x)) ;
```

$$\sin(\sqrt{\lambda} l) v^2 \sqrt{2} \sqrt{\frac{1}{l}} D_1(u)(l, t) - \cos(\sqrt{\lambda} l) \sqrt{\lambda} v^2 \sqrt{2} \sqrt{\frac{1}{l}} u(l, t) + \sqrt{\lambda} v^2 \sqrt{2} \sqrt{\frac{1}{l}} u(0, t) + \int_0^l -\sin(\sqrt{\lambda} x) \lambda v^2 \sqrt{2} \sqrt{\frac{1}{l}} u(x, t) dx$$

Учитываем в полученном выражении граничные условия

> subs ({sin(sqrt(lambda)*l)=0, bound_c[1], bound_c[2]}, %);

$$\int_0^l -\sin(\sqrt{\lambda} x) \lambda v^2 \sqrt{2} \sqrt{\frac{1}{l}} u(x, t) dx$$

Таким образом,

> ode2 := subs (op (1, rhs (ode2)) = %, ode2);

$$ode2 := \frac{d^2}{dt^2} U(t) =$$

$$\int_0^l -\sin(\sqrt{\lambda} x) \lambda v^2 \sqrt{2} \sqrt{\frac{1}{l}} u(x, t) dx + \frac{G \sqrt{2} \sqrt{\frac{1}{l}} (\cos(\sqrt{\lambda} l) - 1)}{\sqrt{\lambda}}$$

Учитывая обозначение для трансформанты, окончательно получаем следующее обыкновенное дифференциальное уравнение для этой трансформанты

> ode2 := subs (op (1, rhs (ode2)) = -v^2 * lambda * U(t), ode2);

$$ode2 := \frac{d^2}{dt^2} U(t) = -v^2 \lambda U(t) + \frac{G \sqrt{2} \sqrt{\frac{1}{l}} (\cos(\sqrt{\lambda} l) - 1)}{\sqrt{\lambda}}$$

Решаем полученное уравнение с учетом начальных условий

> sol_ode2 := dsolve ({ode2, U(0)=0, D(U)(0)=0}, U(t));

sol_ode2 := U(t) =

$$-\frac{\cos(v \sqrt{\lambda} t) G \sqrt{2} \sqrt{\frac{1}{l}} (\cos(\sqrt{\lambda} l) - 1)}{v^2 \lambda^{(3/2)}} + \frac{G \sqrt{2} \sqrt{\frac{1}{l}} (\cos(\sqrt{\lambda} l) - 1)}{v^2 \lambda^{(3/2)}}$$

Следовательно, мы нашли трансформанту Гринберга

> assume (l>0):

U := unapply (simplify (subs (lambda=ev(k), rhs (sol_ode2))), k, t);

$$U := (k\sim, t) \rightarrow -\frac{G \sqrt{2} l^{\sim(5/2)} ((-1)^{k\sim} - 1) \left(\cos\left(\frac{v \pi k\sim t}{l\sim}\right) - 1 \right)}{v^2 \pi^3 k\sim^3}$$

Итак, решение нашей задачи определяется следующим формальным рядом

> 'u(x, t) '=Sum(U(k, t) *X(ev(k), x), k=1..infinity);

$$u(x, t) = \sum_{k=1}^{\infty} \left(- \frac{2 G l^{-2} ((-1)^k - 1) \left(\cos\left(\frac{v \pi k t}{l}\right) - 1 \right) \sin\left(\frac{\pi k x}{l}\right)}{v^2 \pi^3 k^3} \right)$$

Полученную формулу удобно преобразовать, заменив индекс суммирования k на $2m+1$

> subs(k=2*m+1, op(1, rhs(%))); assume(m, posint):

$$- \frac{2 G l^{-2} ((-1)^{(2m+1)} - 1) \left(\cos\left(\frac{v \pi (2m+1) t}{l}\right) - 1 \right) \sin\left(\frac{\pi (2m+1) x}{l}\right)}{v^2 \pi^3 (2m+1)^3}$$

> simplify(%): factor(%);

$$\frac{4 G l^{-2} \left(\cos\left(\frac{v \pi (2m+1) t}{l}\right) - 1 \right) \sin\left(\frac{\pi (2m+1) x}{l}\right)}{v^2 \pi^3 (2m+1)^3}$$

> 'u(x, t) '=Sum(%, m=0..infinity);

$$u(x, t) = \sum_{m=0}^{\infty} \left(\frac{4 G l^{-2} \left(\cos\left(\frac{v \pi (2m+1) t}{l}\right) - 1 \right) \sin\left(\frac{\pi (2m+1) x}{l}\right)}{v^2 \pi^3 (2m+1)^3} \right)$$

Окончательно получаем решение в форме следующего ряда

$$u(x, t) = \frac{4Gl^2}{\pi^3 v^2} \sum_{m=0}^{\infty} \left[\frac{\cos\left(\frac{(2m+1)\pi vt}{l}\right) - 1}{(2m+1)^3} \right] \sin\left(\frac{(2m+1)\pi x}{l}\right).$$

Найденное решение, так же как и в задаче 5.1, можно преобразовать к другому виду, если просуммировать часть полученного выражения. А именно, покажем, что справедлива формула

$$\frac{Gx(x-l)}{2v^2} = - \frac{4Gl^2}{\pi^3 v^2} \sum_{k=0}^{\infty} \frac{\sin\left(\frac{(2k+1)\pi x}{2l}\right)}{(2k+1)^3}.$$

Эта формула легко устанавливается разложением функции, стоящей слева, в ряд Фурье по собственным функциям нашей задачи Штурма-Лиувилля. Действительно, определим эту функцию в Maple

> g:=x->G*x*(x-l)/2/v^2;

$$g := x \rightarrow \frac{1}{2} \frac{G x (x - l)}{v^2}$$

Коэффициенты разложения определяются по формуле

> Int (g (x) *X (ev (k) , x) , x=0 .. l) ;

$$\int_0^{l\sim} \frac{1}{2} \frac{G x (x - l\sim) \sqrt{2} \sin\left(\frac{\pi k\sim x}{l\sim}\right)}{v^2 \sqrt{l\sim}} dx$$

> factor (value (%)) ; C:=unapply (simplify (subs (k=2*m+1, %)) , m) ;

$$\frac{l\sim^{(5/2)} G \sqrt{2} ((-1)^{k\sim} - 1)}{\pi^3 k\sim^3 v^2}$$

$$C := m\sim \rightarrow -\frac{2 l\sim^{(5/2)} G \sqrt{2}}{\pi^3 (2 m\sim + 1)^3 v^2}$$

Таким образом, мы показали, что справедлива формула

> g (x) =Sum (C (m) *X (ev (2*m+1) , x) , m=0 .. infinity) ;

$$\frac{G x (x - l\sim)}{2 v^2} = \sum_{m\sim=0}^{\infty} \left(-\frac{4 l\sim^2 G \sin\left(\frac{\pi (2 m\sim + 1) x}{l\sim}\right)}{\pi^3 (2 m\sim + 1)^3 v^2} \right)$$

Следовательно, решение задачи можно представить и в таком виде

$$u(x, t) = \frac{Gx(x-l)}{2v^2} + \frac{4Gl^2}{\pi^3 v^2} \sum_{k=0}^{\infty} \frac{\cos\left(\frac{(2k+1)\pi vt}{l}\right)}{(2k+1)^3} \sin\left(\frac{(2k+1)\pi x}{l}\right).$$

Как и в задаче 5.1, отметим, что вторая форма решения сразу получается, если искать это решение методом сведения задачи к однородной задаче, то есть в форме

$$u(x, t) = u_0(x) + w(x, t),$$

где $u_0(x)$ — стационарное решение следующей задачи

$$v^2 \frac{d^2 u_0}{dx^2} - G = 0;$$

$$u_0|_{x=0} = 0, \quad u_0|_{x=l} = 0.$$

Легко видеть, что

$$u_0(x) = \frac{Gx(x-l)}{2v^2}.$$

Тогда для определения функции $w(x, t)$ будем иметь следующую однородную задачу

$$\frac{\partial^2 w}{\partial t^2} = v^2 \frac{\partial^2 w}{\partial x^2};$$

$$w|_{t=0} = -u_0(x), \quad \frac{\partial w}{\partial t}\bigg|_{t=0} = 0; \quad w|_{x=0} = 0, \quad w|_{x=l} = 0,$$

которая легко решается методом Фурье

$$w(x, t) = \frac{4Gl^2}{\pi^3 v^2} \sum_{k=1}^{\infty} \frac{\cos\left(\frac{(2k+1)\pi vt}{l}\right)}{(2k+1)^3} \sin\left(\frac{(2k+1)\pi x}{l}\right).$$

Предоставляем читателю самостоятельно проделать все необходимые выкладки в Maple и убедиться в сказанном.

Задача 5.3. Решить смешанную задачу для неоднородного уравнения гиперболического типа

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + 2t, \quad 0 < x < 1, \quad t > 0;$$

$$u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}\bigg|_{t=0} = x; \quad u|_{x=0} = 0, \quad \frac{\partial u}{\partial x}\bigg|_{x=1} = t.$$

Решение. Определяем задачу в Maple

```
> eq:=diff(u(x,t),t$2)=diff(u(x,t),x$2)+2*t; 0<x,x<1,t>0;
```

$$eq := \frac{\partial^2}{\partial t^2} u(x, t) = \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) + 2 t$$

$$0 < x, x < 1, 0 < t$$

```
> init_c:=[u(x,0)=0,D[2](u)(x,0)=x];
```

$$init_c := [u(x, 0) = 0, D_2(u)(x, 0) = x]$$

```
> bound_c:=[u(0,t)=0,D[1](u)(1,t)=t];
```

$$bound_c := [u(0, t) = 0, D_1(u)(1, t) = t]$$

Будем искать решение задачи в форме $u(x, t) = xt + w(x, t)$. Определим эту функцию в Maple и подставим ее в исходное уравнение. Получим

```
> u:=(x,t)->x*t+w(x,t);
```

$$u := (x, t) \rightarrow x t + w(x, t)$$

```
> eq1:=eq;
```

$$eq1 := \frac{\partial^2}{\partial t^2} w(x, t) = \left(\frac{\partial^2}{\partial x^2} w(x, t) \right) + 2 t$$

Преобразуем соответствующие дополнительные условия — начальные

```
> init_c1:=init_c;
```

$$init_c1 := [init_c1[1], lhs(init_c1[2]) - x = 0];$$

$$\text{init_c1} := [w(x, 0) = 0, x + D_2(w)(x, 0) = x]$$

$$\text{init_c1} := [w(x, 0) = 0, D_2(w)(x, 0) = 0]$$

и граничные условия

$$> \text{bound_c1} := \text{bound_c};$$

$$\text{bound_c1} := [\text{bound_c1}[1], \text{lhs}(\text{bound_c1}[2]) - t = 0];$$

$$\text{bound_c1} := [w(0, t) = 0, t + D_1(w)(1, t) = t]$$

$$\text{bound_c1} := [w(0, t) = 0, D_1(w)(1, t) = 0]$$

Итак, для определения функции $w(x, t)$ будем иметь следующую задачу

$$\frac{\partial^2 w}{\partial t^2} = \frac{\partial^2 w}{\partial x^2} + 2t, \quad 0 < x < 1, \quad t > 0;$$

$$w|_{t=0} = 0, \quad \frac{\partial w}{\partial t}|_{t=0} = 0; \quad w|_{x=0} = 0, \quad \frac{\partial w}{\partial x}|_{x=1} = 0.$$

Это — задача с однородными условиями и неоднородным уравнением. Решаем ее методом Гринберга. Для этого производим разделение переменных в соответствующем однородном уравнении

$$> \text{pdsolve}(\text{subs}(2 * t = 0, \text{eq1}), \text{HINT} = X(x) * T(t));$$

$$(w(x, t) = X(x) T(t)) \&\text{where} \left[\left\{ \frac{d^2}{dx^2} X(x) = -c_1 X(x), \frac{d^2}{dt^2} T(t) = -c_1 T(t) \right\} \right]$$

Формулируем соответствующую задачу Штурма-Лиувилля

$$> \text{ode1} := \text{subs}(_c[1] = -\lambda, \text{op}([2, 1, 1], \%));$$

$$\text{bound} := [X(0) = 0, D(X)(1) = 0];$$

$$\text{ode1} := \frac{d^2}{dx^2} X(x) = -\lambda X(x)$$

$$\text{bound} := [X(0) = 0, D(X)(1) = 0]$$

Определяем собственные числа этой задачи

$$> \text{sol_ode} := \text{dsolve}(\text{ode1}, X(x));$$

$$\text{sol_ode} := X(x) = _C1 \sin(\sqrt{\lambda} x) + _C2 \cos(\sqrt{\lambda} x)$$

$$> X := \text{unapply}(\text{rhs}(\text{sol_ode}), x);$$

$$X := x \rightarrow _C1 \sin(\sqrt{\lambda} x) + _C2 \cos(\sqrt{\lambda} x)$$

$$> \text{sist} := \{\text{bound}[1], \text{bound}[2]\};$$

$$\text{sist} := \{ _C2 = 0, _C1 \cos(\sqrt{\lambda}) \sqrt{\lambda} - _C2 \sin(\sqrt{\lambda}) \sqrt{\lambda} = 0 \}$$

$$> A := \text{linalg}[\text{genmatrix}](\text{sist}, \{ _C1, _C2 \});$$

$$A := \begin{bmatrix} 0 & 1 \\ \cos(\sqrt{\lambda}) \sqrt{\lambda} & -\sin(\sqrt{\lambda}) \sqrt{\lambda} \end{bmatrix}$$

> Delta:=linalg[det](A);

$$\Delta := -\cos(\sqrt{\lambda}) \sqrt{\lambda}$$

> _EnvAllSolutions:=true:

solve(simplify(Delta/sqrt(lambda))=0, lambda);

$$\frac{\pi^2 (1 + 2 _Z1\sim)^2}{4}$$

В результате мы получили счетное множество собственных значений

$\lambda = \lambda_k = \left(\frac{(2k+1)\pi}{2} \right)^2$, $k = 0, 1, 2, \dots$. Следовательно, спектр задачи — дискретный.

Определим найденные собственные значения в Maple как функцию целочисленного аргумента

> ev:=unapply(subs(_Z1=k, %), k);

$$ev := k \rightarrow \frac{1}{4} \pi^2 (1 + 2k)^2$$

Найдем теперь соответствующие собственные функции

> X:='X':assume(k, integer):

subs(lambda=ev(k), odel);

dsolve({%, bound[1], bound[2]}, X(x));

$$\frac{d^2}{dx^2} X(x) = -\frac{1}{4} \pi^2 (1 + 2k)^2 X(x)$$

$$X(x) = _C1 \sin\left(\left(\frac{1}{2} \pi + \pi k\right) x\right)$$

Сразу же и нормируем их

> rhs(%)/_C1;

X:=unapply(%/simplify(int(%^2, x=0..1)^(1/2)), k, x);

$$\sin\left(\left(\frac{1}{2} \pi + \pi k\right) x\right)$$

$$X := (k, x) \rightarrow \sin\left(\left(\frac{\pi}{2} + \pi k\right) x\right) \sqrt{2}$$

Итак, нормированные собственные функции будут $X_k(x) = \sqrt{2} \sin\left(\frac{(2k+1)\pi x}{2}\right)$.

Удобно переопределить их в Maple следующим образом

```
> X:=(lambda, x)->sqrt(2)*sin(sqrt(lambda)*x);
      X := (λ, x) → √2 sin(√λ x)
```

Умножаем теперь исходное уравнение на собственные функции и интегрируем по отрезку [0,1]. Получим

```
> ode2:=Int(lhs(eq1)*X(lambda, x), x=0..1)=
      Int(op(1, rhs(eq1))*X(lambda, x), x=0..1)+
      int(op(2, rhs(eq1))*X(lambda, x), x=0..1);
```

$$ode2 := \int_0^1 \left(\frac{\partial^2 w(x, t)}{\partial t^2} \right) \sqrt{2} \sin(\sqrt{\lambda} x) dx = \int_0^1 \left(\frac{\partial^2 w(x, t)}{\partial x^2} \right) \sqrt{2} \sin(\sqrt{\lambda} x) dx - \frac{2 t \sqrt{2} (\cos(\sqrt{\lambda}) - 1)}{\sqrt{\lambda}}$$

Обозначим трансформанту Гринберга

$$w_k(t) = \int_0^1 w(x, t) X_k(x) dx.$$

Тогда (здесь для простоты обозначено $w_k(t) = W(t)$)

```
> ode2:=
      subs({lhs(ode2)=diff(W(t), t$2), cos(sqrt(lambda))=0}, ode2);
```

$$ode2 := \frac{d^2}{dt^2} W(t) = \int_0^1 \left(\frac{\partial^2 w(x, t)}{\partial x^2} \right) \sqrt{2} \sin(\sqrt{\lambda} x) dx + \frac{2 t \sqrt{2}}{\sqrt{\lambda}}$$

Как обычно, преобразовываем интеграл два раза по частям

```
> op(1, rhs(ode2));
```

$$\int_0^1 \left(\frac{\partial^2 w(x, t)}{\partial x^2} \right) \sqrt{2} \sin(\sqrt{\lambda} x) dx$$

```
> student[intparts](%, sin(sqrt(lambda)*x));
```

$$\sin(\sqrt{\lambda}) \sqrt{2} D_1(w)(1, t) - \int_0^1 \cos(\sqrt{\lambda} x) \sqrt{\lambda} \sqrt{2} \left(\frac{\partial}{\partial x} w(x, t) \right) dx$$

```
> student[intparts](%, cos(sqrt(lambda)*x));
```

$$\sin(\sqrt{\lambda}) \sqrt{2} D_1(w)(1, t) - \cos(\sqrt{\lambda}) \sqrt{\lambda} \sqrt{2} w(1, t) + \sqrt{\lambda} \sqrt{2} w(0, t) + \int_0^1 -\sin(\sqrt{\lambda} x) \lambda \sqrt{2} w(x, t) dx$$

С учетом граничных условий находим

> subs({cos(sqrt(lambda))=0, bound_c1[1], bound_c1[2]}, %);

$$\int_0^1 -\sin(\sqrt{\lambda} x) \lambda \sqrt{2} w(x, t) dx$$

Итак, мы получили

> ode2:=subs(op(1, rhs(ode2))=%, ode2);

$$ode2 := \frac{d^2}{dt^2} W(t) = \int_0^1 -\sin(\sqrt{\lambda} x) \lambda \sqrt{2} w(x, t) dx + \frac{2 t \sqrt{2}}{\sqrt{\lambda}}$$

или, окончательно с учетом обозначения для трансформанты

> ode2:=subs(op(1, rhs(ode2))=-lambda*W(t), ode2);

$$ode2 := \frac{d^2}{dt^2} W(t) = -\lambda W(t) + \frac{2 t \sqrt{2}}{\sqrt{\lambda}}$$

Решаем полученное уравнение с учетом однородных начальных условий

> sol_ode2:=dsolve({ode2, W(0)=0, D(W)(0)=0}, W(t));

$$sol_ode2 := W(t) = -\frac{2 \sin(\sqrt{\lambda} t) \sqrt{2}}{\lambda^2} + \frac{2 \sqrt{2} t}{\lambda^{(3/2)}}$$

Подставим теперь вместо λ собственные значения и упростим полученное решение

> assume(k, posint);

simplify(subs(lambda=ev(k), rhs(sol_ode2))); collect(%, t);

$$\frac{16 \left(2 t \pi k - 2 \sin\left(\frac{\pi (1 + 2 k) t}{2}\right) + t \pi \right) \sqrt{2}}{\pi^4 (1 + 2 k)^4}$$

$$\frac{16 (2 \pi k + \pi) \sqrt{2} t}{\pi^4 (1 + 2 k)^4} - \frac{32 \sin\left(\frac{\pi (1 + 2 k) t}{2}\right) \sqrt{2}}{\pi^4 (1 + 2 k)^4}$$

> normal(op(1, %)) + op(2, %);

$$\frac{16 t \sqrt{2}}{\pi^3 (1 + 2 k)^3} - \frac{32 \sin\left(\frac{\pi (1 + 2 k) t}{2}\right) \sqrt{2}}{\pi^4 (1 + 2 k)^4}$$

Таким образом, мы окончательно получили трансформанту Гринберга в виде

$$w_k(t) = \frac{16\sqrt{2}t}{\pi^3(2k+1)^3} - \frac{32\sqrt{2} \sin\left(\frac{(2k+1)\pi t}{2}\right)}{\pi^4(2k+1)^4}$$

или в Maple

> W:=unapply(% , k, t);

$$W := (k\sim, t) \rightarrow \frac{16 t \sqrt{2}}{\pi^3 (1 + 2 k\sim)^3} - \frac{32 \sin\left(\frac{1}{2} \pi (1 + 2 k\sim) t\right) \sqrt{2}}{\pi^4 (1 + 2 k\sim)^4}$$

Выпишем теперь решение нашей задачи

> 'w(x, t)' = Sum(W(k, t) * X(ev(k), x), k=0..infinity);

w(x, t) =

$$\sum_{k\sim=0}^{\infty} \left(\frac{16 t \sqrt{2}}{\pi^3 (1 + 2 k\sim)^3} - \frac{32 \sin\left(\frac{\pi (1 + 2 k\sim) t}{2}\right) \sqrt{2}}{\pi^4 (1 + 2 k\sim)^4} \right) \sqrt{2} \sin\left(\frac{\pi (1 + 2 k\sim) x}{2}\right)$$

То есть

$$w(x, t) = \sum_{k=0}^{\infty} \left[\frac{32t}{\pi^3(2k+1)^3} - \frac{64 \sin\left(\frac{(2k+1)\pi t}{2}\right)}{\pi^4(2k+1)^4} \right] \sin\left(\frac{(2k+1)\pi x}{2}\right).$$

Здесь, так же, как и в предыдущей задаче, можно просуммировать часть ряда. А именно, покажем, что функция

> f:=x->x*(2-x);

$$f := x \rightarrow x(2-x)$$

является суммой ряда

> Sum(32/Pi^3/(1+2*k)^3*sin(Pi*(1+2*k)*x/2), k=0..infinity);

$$\sum_{k=0}^{\infty} \left(\frac{32 \sin\left(\frac{\pi (1 + 2 k) x}{2}\right)}{\pi^3 (1 + 2 k)^3} \right)$$

то есть, справедлива формула

$$x(2-x) = \sum_{k=0}^{\infty} \frac{32 \sin\left(\frac{(2k+1)\pi x}{2}\right)}{\pi^3(2k+1)^3}.$$

Эта формула легко устанавливается разложением функции, стоящей слева, в ряд Фурье по собственным функциям задачи Штурма-Лиувилля. Очевидно, коэффициенты разложения будут

> Int (f (x) *X (ev (k) , x) , x=0..1) =
factor (int (f (x) *X (ev (k) , x) , x=0..1));

$$\int_0^1 x(2-x) \sqrt{2} \sin\left(\frac{\pi(1+2k)x}{2}\right) dx = \frac{16\sqrt{2}}{\pi^3(1+2k)^3}$$

и, таким образом, мы показали, что

> f (x) =Sum (rhs (%) *X (ev (k) , x) , k=0..infinity) ;

$$x(2-x) = \sum_{k=0}^{\infty} \left(\frac{32 \sin\left(\frac{\pi(1+2k)x}{2}\right)}{\pi^3(1+2k)^3} \right)$$

Учитывая сказанное, решение задачи можно записать так

> 'u (x, t) ' =factor (x*t+t*f (x)) -
Sum ((64*sin (Pi*(1+2*k)*t/2) /Pi^4/(1+2*k)^4) *
sin (Pi*(1+2*k)*x/2) , k=0..infinity) ;

$$u(x, t) = -x t (-3 + x) - \left(\sum_{k=0}^{\infty} \left(\frac{64 \sin\left(\frac{\pi(1+2k)t}{2}\right) \sin\left(\frac{\pi(1+2k)x}{2}\right)}{\pi^4(1+2k)^4} \right) \right)$$

Итак, мы получили формальное решение

$$u(x, t) = xt(3-x) - \sum_{k=0}^{\infty} \frac{64 \sin\left(\frac{(2k+1)\pi t}{2}\right)}{\pi^4(2k+1)^4} \sin\left(\frac{(2k+1)\pi x}{2}\right).$$

Задача 5.4. Решить смешанную задачу для неоднородного уравнения параболического типа

$$\frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2} + t(x+1), \quad 0 < x < 1, \quad t > 0;$$

$$u|_{t=0} = 0, \quad \frac{\partial u}{\partial x}|_{x=0} = t^2, \quad u|_{x=1} = t^2.$$

Решение. Определяем задачу в Maple

> eq:=diff (u (x, t) , t) =diff (u (x, t) , x\$2) +t*(x+1) ; 0<x, x<1, t>0;

$$eq := \frac{\partial}{\partial t} u(x, t) = \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) + t(x+1)$$

$$0 < x, x < 1, 0 < t$$

> init_c:=u (x, 0) =0;

$$init_c := u(x, 0) = 0$$

```
> bound_c := [u(1, t) = t^2, D[1](u)(0, t) = t^2];
      bound_c := [u(1, t) = t^2, D_1(u)(0, t) = t^2]
```

Здесь, так же, как и в задаче 5.3, удобно избавиться от неоднородности в граничных условиях. Для этого положим $u(x, t) = xt^2 + w(x, t)$. Тогда для функции w получим следующую задачу

```
> u := (x, t) -> x*t^2 + w(x, t);
      u := (x, t) -> x t^2 + w(x, t)
> eq1 := collect(lhs(eq) - rhs(eq), t) = 0;
      eq1 := (x - 1) t + \left(\frac{\partial}{\partial t} w(x, t)\right) - \left(\frac{\partial^2}{\partial x^2} w(x, t)\right) = 0
> init_c1 := init_c;
      init_c1 := w(x, 0) = 0
> bound_c1 := bound_c;
bound_c1 := [lhs(bound_c1[1]) - t^2 = 0, lhs(bound_c1[2]) - t^2 = 0];
      bound_c1 := [t^2 + w(1, t) = t^2, t^2 + D_1(w)(0, t) = t^2]
      bound_c1 := [w(1, t) = 0, D_1(w)(0, t) = 0]
```

то есть

$$\frac{\partial w}{\partial t} - \frac{\partial^2 w}{\partial x^2} + t(x-1), \quad 0 < x < 1, \quad t > 0;$$

$$w|_{t=0} = 0, \quad \frac{\partial w}{\partial x}\Big|_{x=0} = 0, \quad w|_{x=1} = 0.$$

Эту задачу решаем методом Гринберга. Производим разделение переменных в однородном уравнении

```
> pdsolve(subs((x-1)*t=0, eq1), HINT=X(x)*T(t));
      (w(x, t) = X(x) T(t)) &where \left[ \left\{ \frac{d}{dt} T(t) = \_c_1 T(t), \frac{d^2}{dx^2} X(x) = \_c_1 X(x) \right\} \right]
```

Формулируем соответствующую задачу Штурма-Лиувилля

```
> ode1 := subs(_c[1] = -lambda, op([2, 1, 1], %));
      bound := [X(1) = 0, D(X)(0) = 0];
      ode1 := \frac{d^2}{dx^2} X(x) = -\lambda X(x)
      bound := [X(1) = 0, D(X)(0) = 0]
```

Находим собственные значения

```

> sol_ode:=dsolve(ode1,X(x));
      sol_ode := X(x) = _C1 sin(√λ x) + _C2 cos(√λ x)
> X:=unapply(rhs(sol_ode),x);
      X := x → _C1 sin(√λ x) + _C2 cos(√λ x)
> sist:={bound[1],bound[2]};
      sist := {_C1 sin(√λ) + _C2 cos(√λ) = 0, _C1 √λ = 0}
> A:=linalg[genmatrix](sist, {_C1, _C2});
      A := [ sin(√λ)  cos(√λ)
            [ √λ      0
> Delta:=linalg[det](A);
      Δ := -cos(√λ) √λ
> _EnvAllSolutions:=true:
      solve(simplify(Delta/sqrt(lambda))=0,lambda);
      π² (1 + 2 _Z1~)²
      4
> ev:=unapply(subs(_Z1=k,%),k);
      ev := k → ¼ π² (1 + 2 k)²

```

Итак, мы получили дискретный спектр собственных значений $\lambda = \lambda_k = \left(\frac{(2k+1)\pi}{2}\right)^2$,

$k = 0, 1, 2, \dots$. Читателю мы рекомендуем самостоятельно убедиться в том, что $\lambda = 0$ не является собственным значением рассматриваемой задачи.

Находим теперь собственные функции

```

> X:='X':assume(k,integer):
      subs(lambda=ev(k),ode1);
      dsolve({%,bound[1],bound[2]},X(x));
      d²
      dx² X(x) = -¼ π² (1 + 2 k~)² X(x)
      X(x) = _C2 cos( (½ π + π k~) x )

```

Сразу же и нормируем их

```

> rhs(%)/_C2;
      X:=unapply(%/simplify(int(^(2,x=0..1)^(1/2)),k,x);

```

$$\cos\left(\left(\frac{1}{2}\pi + \pi k\right)x\right)$$

$$X := (k, x) \rightarrow \cos\left(\left(\frac{\pi}{2} + \pi k\right)x\right)\sqrt{2}$$

Нормированные собственные функции будут

$$X_k(x) = \sqrt{2} \cos(\sqrt{\lambda_k}x) = \sqrt{2} \cos\left(\frac{(2k+1)\pi x}{2}\right).$$

Определим их в Maple как

```
> X := (lambda, x) -> sqrt(2) * cos(sqrt(lambda) * x);
      X := (lambda, x) -> sqrt(2) * cos(sqrt(lambda) * x)
```

Умножаем теперь исходное уравнение на собственные функции и интегрируем по отрезку [0,1]. Получим

```
> ode2 := int(op(1, lhs(eq1)) * X(lambda, x), x=0..1) +
      int(op(2, lhs(eq1)) * X(lambda, x), x=0..1) +
      int(op(3, lhs(eq1)) * X(lambda, x), x=0..1);
```

$$\text{ode2} := \frac{t\sqrt{2}(\cos(\sqrt{\lambda}) - 1)}{\lambda} + \int_0^1 \left(\frac{\partial}{\partial t} w(x, t)\right) \sqrt{2} \cos(\sqrt{\lambda} x) dx$$

$$+ \int_0^1 -\left(\frac{\partial^2}{\partial x^2} w(x, t)\right) \sqrt{2} \cos(\sqrt{\lambda} x) dx$$

Обозначим трансформанту Гринберга

$$w_k(t) = \int_0^1 w(x, t) X_k(x) dx.$$

Тогда (как и ранее, для простоты записи принимаем $w_k(t) = W(t)$)

```
> ode2 := subs({op(2, ode2) = diff(W(t), t), cos(sqrt(lambda)) = 0},
      ode2);
```

$$\text{ode2} := -\frac{t\sqrt{2}}{\lambda} + \left(\frac{d}{dt} W(t)\right) + \int_0^1 -\left(\frac{\partial^2}{\partial x^2} w(x, t)\right) \sqrt{2} \cos(\sqrt{\lambda} x) dx$$

Интеграл преобразовываем два раза по частям

```
> op(3, ode2);
```


$$\int_0^1 -\left(\frac{\partial^2}{\partial x^2} w(x, t)\right) \sqrt{2} \cos(\sqrt{\lambda} x) dx$$

> student[intparts](%, cos(sqrt(lambda)*x));

$$-\cos(\sqrt{\lambda}) \sqrt{2} D_1(w)(1, t) + \sqrt{2} D_1(w)(0, t) - \int_0^1 \sin(\sqrt{\lambda} x) \sqrt{\lambda} \sqrt{2} \left(\frac{\partial}{\partial x} w(x, t)\right) dx$$

> student[intparts](%, sin(sqrt(lambda)*x));

$$-\cos(\sqrt{\lambda}) \sqrt{2} D_1(w)(1, t) + \sqrt{2} D_1(w)(0, t) - \sin(\sqrt{\lambda}) \sqrt{\lambda} \sqrt{2} w(1, t) + \int_0^1 \cos(\sqrt{\lambda} x) \lambda \sqrt{2} w(x, t) dx$$

Таким образом, с учетом граничных условий, получаем

> subs({cos(sqrt(lambda))=0, bound_c1[1], bound_c1[2]}, %);

$$\int_0^1 \cos(\sqrt{\lambda} x) \lambda \sqrt{2} w(x, t) dx$$

То есть, будем иметь

> ode2:=subs(op(3, ode2)=%, ode2);

$$ode2 := -\frac{t\sqrt{2}}{\lambda} + \left(\frac{d}{dt} W(t)\right) + \int_0^1 \cos(\sqrt{\lambda} x) \lambda \sqrt{2} w(x, t) dx$$

или, с учетом обозначения для трансформанты

> ode2:=subs(op(3, ode2)=lambda*W(t), ode2);

$$ode2 := -\frac{t\sqrt{2}}{\lambda} + \left(\frac{d}{dt} W(t)\right) + \lambda W(t)$$

Решаем полученное уравнение с учетом начальных условий

> sol_ode2:=dsolve({ode2, W(0)=0}, W(t));

$$sol_ode2 := W(t) = \frac{\sqrt{2} t}{\lambda^2} - \frac{\sqrt{2}}{\lambda^3} + \frac{e^{(-\lambda t)} \sqrt{2}}{\lambda^3}$$

Таким образом, мы нашли трансформанту Гринберга

$$w_k(t) = \sqrt{2} \left[\frac{t^2}{\lambda_k^2} - \frac{1}{\lambda_k^3} + \frac{\sqrt{2} e^{-\lambda_k t}}{\lambda_k^3} \right].$$

Выпишем теперь формальное решение задачи в виде ряда

$$w(x, t) = \sum_{k=0}^{\infty} w_k(t) X_k(x).$$

Подставим вместо параметра λ найденные собственные значения и упростим

> assume(k, posint) :

simplify(subs(lambda=ev(k), rhs(sol_ode2))); collect(%, t) ;

$$\frac{16 \sqrt{2} \left(\pi^2 t + 4 \pi^2 t k + 4 \pi^2 t k^2 - 4 + 4 e^{\left(-\frac{\pi^2 (1+2k)^2 t}{4} \right)} \right)}{\pi^6 (1+2k)^6}$$

$$\frac{16 \sqrt{2} (\pi^2 + 4 \pi^2 k + 4 \pi^2 k^2) t}{\pi^6 (1+2k)^6} + \frac{16 \sqrt{2} \left(-4 + 4 e^{\left(-\frac{\pi^2 (1+2k)^2 t}{4} \right)} \right)}{\pi^6 (1+2k)^6}$$

> normal(op(1, %)) + factor(op(2, %)) ;

$$\frac{16 \sqrt{2} t}{\pi^4 (1+2k)^4} + \frac{64 \sqrt{2} \left(-1 + e^{\left(-\frac{\pi^2 (1+2k)^2 t}{4} \right)} \right)}{\pi^6 (1+2k)^6}$$

Определим теперь окончательно трансформанту Гринберга в Maple

> W:=unapply(%, k, t) ;

$$W := (k, t) \rightarrow \frac{16 \sqrt{2} t}{\pi^4 (1+2k)^4} + \frac{64 \sqrt{2} (-1 + e^{(-1/4 \pi^2 (1+2k)^2 t})}){\pi^6 (1+2k)^6}$$

Таким образом, решение задачи имеет вид

> 'w(x, t)' = Sum(W(k, t) * X(ev(k), x), k=0..infinity) ;

$$w(x, t) = \sum_{k=0}^{\infty}$$

$$\left(\frac{16 \sqrt{2} t}{\pi^4 (1+2k)^4} + \frac{64 \sqrt{2} \left(-1 + e^{\left(-\frac{\pi^2 (1+2k)^2 t}{4} \right)} \right)}{\pi^6 (1+2k)^6} \right) \sqrt{2} \cos\left(\frac{\pi (1+2k) x}{2} \right)$$

А тогда искомая функция определяется по формуле

> w:=(x, t) -> Sum(W(k, t) * X(ev(k), x), k=0..infinity) ;

'u(x, t)' = u(x, t) ;

$$u(x, t) = x t^2 + \left(\sum_{k=0}^{\infty} \left(\frac{16 \sqrt{2} t}{\pi^4 (1+2k)^4} + \frac{64 \sqrt{2} \left(-1 + e^{\left(-\frac{\pi^2 (1+2k)^2 t}{4} \right)} \right)}{\pi^6 (1+2k)^6} \right) \sqrt{2} \cos\left(\frac{\pi (1+2k) x}{2} \right) \right)$$

Окончательно мы получили формальное решение задачи в виде следующего ряда

$$u(x, t) = xt^2 + \sum_{k=0}^{\infty} \left\{ \frac{32t}{\pi^4 (2k+1)^4} - \frac{128 \left[1 - e^{-\frac{\pi^2 (2k+1)^2 t}{4}} \right]}{\pi^6 (2k+1)^6} \right\} \cos\left(\frac{\pi(2k+1)x}{2}\right).$$

Задача 5.5. Решить следующую краевую задачу в полукруге радиуса b :

$$\frac{1}{a^2} \frac{\partial^2 u}{\partial t^2} = \Delta u + \rho \cos(\varphi) \sin(\omega t), \quad 0 \leq \rho \leq b, \quad 0 \leq \varphi \leq \pi;$$

$$u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}|_{t=0} = 0, \quad \frac{\partial u}{\partial n}|_{\Gamma} = 0.$$

Решение. Очевидно, задачу удобно решать в полярных координатах. Сформулируем задачу: найти функцию $u(\rho, \varphi, t)$, удовлетворяющую уравнению

$$\frac{1}{a^2} \frac{\partial^2 u}{\partial t^2} = \frac{1}{\rho} \frac{\partial}{\partial \rho} \left(\rho \frac{\partial u}{\partial \rho} \right) + \frac{1}{\rho^2} \frac{\partial^2 u}{\partial \varphi^2} + \rho \cos(\varphi) \sin(\omega t), \quad 0 \leq \rho \leq b, \quad 0 \leq \varphi \leq \pi;$$

начальным условиям

$$u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}|_{t=0} = 0$$

и граничным условиям

$$\frac{\partial u}{\partial \rho}|_{\rho=b} = 0, \quad \frac{\partial u}{\partial \varphi}|_{\varphi=0} = 0, \quad \frac{\partial u}{\partial \varphi}|_{\varphi=\pi} = 0.$$

Определим задачу в Maple

```
> PDE:=1/a^2*diff(u(rho,phi,t),t$2)=
VectorCalculus[Laplacian](u(rho,phi,t),'polar'[rho,phi])+
rho*cos(phi)*sin(omega*t);
bound:=D[1](u)(b,phi,t)=0;
init:=[u(rho,phi,0)=0,D[3](u)(rho,phi,0)=0];
```

$$PDE := \frac{\frac{\partial^2}{\partial t^2} u(\rho, \phi, t)}{a^2} =$$

$$\frac{\left(\frac{\partial}{\partial \rho} u(\rho, \phi, t) \right) + \rho \left(\frac{\partial^2}{\partial \rho^2} u(\rho, \phi, t) \right) + \frac{\frac{\partial^2}{\partial \phi^2} u(\rho, \phi, t)}{\rho}}{\rho} + \rho \cos(\phi) \sin(\omega t)$$

$$bound := D_1(u)(b, \phi, t) = 0$$

$$init := [u(\rho, \phi, 0) = 0, D_3(u)(\rho, \phi, 0) = 0]$$

Будем искать решение задачи в виде $u(\rho, \phi, t) = \cos(\phi)v(\rho, t)$. Тогда, очевидно, так выбранная функция $u(\rho, \phi, t)$ заведомо удовлетворяет граничным условиям по координате ϕ

$$\left. \frac{\partial u}{\partial \phi} \right|_{\phi=0, \pi} = -\sin(\phi)v(\rho, t) \Big|_{\phi=0, \pi} = 0.$$

Подставим эту функцию в исходное уравнение и получим уравнение для определения функции $v(\rho, t)$. Проведем все выкладки в Maple. Имеем

```
> u := (rho, phi, t) -> cos(phi) * v(rho, t);
      u := (rho, phi, t) -> cos(phi) v(rho, t)

> PDE1 :=
expand(collect(lhs(PDE), cos(phi))/cos(phi)) =
expand(collect(rhs(PDE), cos(phi))/cos(phi));

PDE1 := \frac{\frac{\partial^2}{\partial t^2} v(\rho, t)}{a^2} = \frac{\frac{\partial}{\partial \rho} v(\rho, t)}{\rho} + \left( \frac{\partial^2}{\partial \rho^2} v(\rho, t) \right) - \frac{v(\rho, t)}{\rho^2} + \rho \sin(\omega t)

> bound; bound_v := bound[1]/cos(phi);
      [cos(phi) D_1(v)(b, t) = 0, 0 = 0, 0 = 0]
      bound_v := D_1(v)(b, t) = 0

> init; init_v := [init[1]/cos(phi), init[2]/cos(phi)];
      [cos(phi) v(rho, 0) = 0, cos(phi) D_2(v)(rho, 0) = 0]
      init_v := [v(rho, 0) = 0, D_2(v)(rho, 0) = 0]
```

Следовательно, для определения функции $v(\rho, t)$ мы получили уравнение

$$\frac{1}{a^2} \frac{\partial^2 v}{\partial t^2} = \frac{1}{\rho} \frac{\partial v}{\partial \rho} + \frac{\partial^2 v}{\partial \rho^2} - \frac{v}{\rho^2} + \rho \sin(\omega t). \quad (*)$$

Соответствующие начальные и граничные условия имеют вид

$$v|_{t=0} = 0, \quad \left. \frac{\partial v}{\partial t} \right|_{t=0} = 0,$$

$$\left. \frac{\partial v}{\partial \rho} \right|_{\rho=b} = 0.$$

Полученную задачу решаем методом Гринберга. Сформулируем соответствующую задачу Штурма-Лиувилля. Для этого произведем разделение переменных в однородном уравнении

```
> PDE0 := lhs (PDE1) = rhs (PDE1) - op (4, rhs (PDE1)) ;
```

$$PDE0 := \frac{\frac{\partial^2}{\partial t^2} v(\rho, t)}{a^2} = \frac{\frac{\partial}{\partial \rho} v(\rho, t)}{\rho} + \left(\frac{\partial^2}{\partial \rho^2} v(\rho, t) \right) - \frac{v(\rho, t)}{\rho^2}$$

```
> spr := pdsolve (PDE0, HINT=R (rho) *T (t)) ;
```

```
spr := (v (rho, t) = R (rho) T (t)) &where
```

$$\left[\left[\left\{ \frac{d^2}{d\rho^2} R(\rho) = -R(\rho) _c_1 + \frac{-\left(\frac{d}{d\rho} R(\rho)\right) \rho + R(\rho)}{\rho^2}, \frac{d^2}{dt^2} T(t) = -T(t) a^2 _c_1 \right\} \right] \right]$$

```
> ode1 :=
```

```
subs (_c [1]=lambda, op (1, op (1, op (2, spr)))) ;
```

$$ode1 := \frac{d^2}{d\rho^2} R(\rho) = -R(\rho) \lambda + \frac{-\left(\frac{d}{d\rho} R(\rho)\right) \rho + R(\rho)}{\rho^2}$$

```
> res := dsolve (ode1) ;
```

$$res := R(\rho) = _C1 \text{BesselJ}(1, \sqrt{\lambda} \rho) + _C2 \text{BesselY}(1, \sqrt{\lambda} \rho)$$

Мы получили однородное дифференциальное уравнение

$$\frac{d^2 R}{d\rho^2} = -\lambda R - \frac{1}{\rho} \frac{dR}{d\rho} - \frac{R}{\rho^2},$$

которое, очевидно, надлежит интегрировать с условиями

$$\left. \frac{dR}{d\rho} \right|_{\rho=b} = 0, \quad R|_{\rho=0} = O(1).$$

Мы сразу же нашли и общее решение этого уравнения, которое имеет вид

$$R(\rho) = C_1 J_1(\sqrt{\lambda}\rho) + C_2 Y_1(\sqrt{\lambda}\rho).$$

В силу ограниченности функции в нуле, полагаем в найденном общем решении $C_2 = 0$. Из другого граничного условия получаем характеристическое уравнение для определения собственных чисел

$$J_1'(\sqrt{\lambda}b) = 0.$$

Или, в Maple

```
> subs(_C2=0, rhs(res));
      _C1 BesselJ(1, sqrt(lambda) rho)
> eq_char := (subs(rho=b, Diff(%/_C1, rho))) = 0;
      eq_char := d/d b BesselJ(1, sqrt(lambda) b) = 0
> eq_char := expand(value(lhs(eq_char)*b)) = 0;
      eq_char := sqrt(lambda) b BesselJ(0, sqrt(lambda) b) - BesselJ(1, sqrt(lambda) b) = 0
```

Видим, что характеристическое уравнение преобразовывается к виду

$$\sqrt{\lambda}b J_0(\sqrt{\lambda}b) - J_1(\sqrt{\lambda}b) = 0.$$

Удобно обозначить $\mu = \sqrt{\lambda}b$. Пусть $\mu_n, n = 1, 2, \dots$ — последовательные положительные корни уравнения $J_1'(\mu) = 0$ или, что то же самое, уравнения

$$\mu J_0(\mu) - J_1(\mu) = 0.$$

Тогда⁴ собственные значения будут $\lambda = \lambda_n = \left(\frac{\mu_n}{b}\right)^2$. Соответствующие собственные функции — $R_n(\rho) = J_1\left(\frac{\mu_n \rho}{b}\right)$.

Или, в Maple: характеристическое уравнение

```
> eq_ch := mu[n]*BesselJ(0, mu[n]) - BesselJ(1, mu[n]) = 0;
      eq_ch := mu_n BesselJ(0, mu_n) - BesselJ(1, mu_n) = 0
```

собственные значения

```
> ev := n -> (mu[n]/b)^2;
```

⁴ Читателю мы предлагаем убедиться в том, что $\lambda = 0$ не является собственным значением задачи.

$$ev := n \rightarrow \frac{\mu_n^2}{b^2}$$

собственные функции

$$> ef := (\rho, n) \rightarrow \text{BesselJ}(1, \mu[n] * \rho / b);$$

$$ef := (\rho, n) \rightarrow \text{BesselJ}\left(1, \frac{\mu_n \rho}{b}\right)$$

Нормируем найденные собственные функции. Вычисляем квадрат нормы

$$> \text{Int}(\rho * ef(\rho, n)^2, \rho = 0..b) =$$

$$\text{int}(\rho * ef(\rho, n)^2, \rho = 0..b);$$

$$\int_0^b \rho \text{BesselJ}\left(1, \frac{\mu_n \rho}{b}\right)^2 d\rho = \frac{1}{2} b^2 (\sqrt{\pi} \mu_n \text{BesselJ}(0, \mu_n)^2 - 2 \sqrt{\pi} \text{BesselJ}(0, \mu_n) \text{BesselJ}(1, \mu_n) + \sqrt{\pi} \mu_n \text{BesselJ}(1, \mu_n)^2) / (\sqrt{\pi} \mu_n)$$

Упростим результат с учетом характеристического уравнения

$$> \text{norm2} := \text{simplify}(\text{subs}(\text{BesselJ}(1, \mu[n]) = \mu[n] * \text{BesselJ}(0, \mu[n]), \text{rhs}(\%)));$$

$$\text{norm2} := \frac{1}{2} b^2 \text{BesselJ}(0, \mu_n)^2 (-1 + \mu_n^2)$$

$$\text{norm2} := \frac{1}{2} b^2 \text{BesselJ}(0, \mu_n)^2 (-1 + \mu_n^2)$$

$$> ef_n := \text{unapply}(ef(\rho, n) / \text{norm2}, \rho, n);$$

$$ef_n := (\rho, n) \rightarrow \frac{2 \text{BesselJ}\left(1, \frac{\mu_n \rho}{b}\right)}{b^2 \text{BesselJ}(0, \mu_n)^2 (-1 + \mu_n^2)}$$

Итак, нормированные собственные функции будут

$$\mathfrak{R}_n(\rho) = \frac{2}{b^2} \frac{J_1\left(\frac{\mu_n \rho}{b}\right)}{(\mu_n^2 - 1) J_0^2(\mu_n)}.$$

Далее, в соответствии с методом Гринберга, ищем решение задачи в форме ряда

$$v(\rho, t) = \sum_{n=1}^{\infty} \frac{R_n(\rho) T_n(t)}{\|R_n\|^2}.$$

Получим уравнение для трансформанты Гринберга

$$T_n(t) = \int_0^b \rho R_n(\rho) v(\rho, t).$$

Для этого умножим уравнение (*) на $\rho R_n(\rho)$ и проинтегрируем по $[0, b]$. Получим

$$\begin{aligned}
 > \text{ode2} := \text{Int}(\text{lhs}(\text{PDE1}) * \rho * R_n(\rho), \rho = 0..b) = \\
 & \text{Int}(\text{op}(1, \text{rhs}(\text{PDE1})) * \rho * R_n(\rho), \rho = 0..b) + \\
 & \text{Int}(\text{op}(2, \text{rhs}(\text{PDE1})) * \rho * R_n(\rho), \rho = 0..b) + \\
 & \text{Int}(\text{op}(3, \text{rhs}(\text{PDE1})) * \rho * R_n(\rho), \rho = 0..b) + \\
 & \text{Int}(\text{op}(4, \text{rhs}(\text{PDE1})) * \rho * \text{ef}(\rho, n), \rho = 0..b); \\
 \\
 \text{ode2} & := \int_0^b \left(\frac{\partial^2}{\partial t^2} v(\rho, t) \right) \frac{\rho R_n(\rho)}{a^2} d\rho = \int_0^b \left(\frac{\partial}{\partial \rho} v(\rho, t) \right) R_n(\rho) d\rho \\
 & + \int_0^b \left(\frac{\partial^2}{\partial \rho^2} v(\rho, t) \right) \rho R_n(\rho) d\rho + \int_0^b - \frac{v(\rho, t) R_n(\rho)}{\rho} d\rho \\
 & + \int_0^b \rho^2 \sin(\omega t) \text{BesselJ}\left(1, \frac{\mu_n \rho}{b}\right) d\rho
 \end{aligned}$$

Упростим последнее выражение справа с учетом характеристического уравнения и введем обозначение трансформанты Гринберга, как $T_n(t) = T(t)$

$$\begin{aligned}
 > \text{ode2} := \text{diff}(T(t), t^2) / a^2 = \text{rhs}(\text{ode2}); \\
 \\
 \text{ode2} & := \frac{d^2}{dt^2} T(t) = \int_0^b \left(\frac{\partial}{\partial \rho} v(\rho, t) \right) R_n(\rho) d\rho + \int_0^b \left(\frac{\partial^2}{\partial \rho^2} v(\rho, t) \right) \rho R_n(\rho) d\rho \\
 & + \int_0^b - \frac{v(\rho, t) R_n(\rho)}{\rho} d\rho + \int_0^b \rho^2 \sin(\omega t) \text{BesselJ}\left(1, \frac{\mu_n \rho}{b}\right) d\rho
 \end{aligned}$$

Итак, мы получили уравнение

$$\begin{aligned}
 \frac{1}{a^2} \frac{d^2 T}{dt^2} & = \int_0^b \frac{\partial v(\rho, t)}{\partial \rho} R_n(\rho) d\rho + \int_0^b \frac{\partial^2 v(\rho, t)}{\partial \rho^2} \rho R_n(\rho) d\rho - \\
 & - \int_0^b \frac{v(\rho, t)}{\rho} R_n(\rho) d\rho + \sin(\omega t) \int_0^b \rho^2 J_1\left(\frac{\mu_n \rho}{b}\right) d\rho. \quad (**)
 \end{aligned}$$

Преобразуем второй интеграл справа два раза по частям. Выполняем все преобразования в Maple. Для этого выделяем интегралы, стоящие в правой части уравнения в отдельные выражения

> _1:=op(1, rhs(ode2)); _2:=op(2, rhs(ode2));
 _3:=op(3, rhs(ode2)); _4:=op(4, rhs(ode2));

$$\begin{aligned} _1 &:= \int_0^b \left(\frac{\partial}{\partial \rho} v(\rho, t) \right) R_n(\rho) d\rho \\ _2 &:= \int_0^b \left(\frac{\partial^2}{\partial \rho^2} v(\rho, t) \right) \rho R_n(\rho) d\rho \\ _3 &:= \int_0^b - \frac{v(\rho, t) R_n(\rho)}{\rho} d\rho \\ _4 &:= \int_0^b \rho^2 \sin(\omega t) \text{BesselJ} \left(1, \frac{\mu_n \rho}{b} \right) d\rho \end{aligned}$$

Преобразовываем второй интеграл по частям

> _2:=student[intparts](_2, rho*R[n](rho));

$$_2 := b R_n(b) \left(\frac{\partial}{\partial b} v(b, t) \right) - \int_0^b \left(R_n(\rho) + \rho \left(\frac{d}{d\rho} R_n(\rho) \right) \right) \left(\frac{\partial}{\partial \rho} v(\rho, t) \right) d\rho$$

То есть, мы получили

$$\int_0^b \frac{\partial^2 v(\rho, t)}{\partial \rho^2} \rho R_n(\rho) d\rho = b R_n(b) \frac{\partial v}{\partial \rho} \Big|_{\rho=b} - \int_0^b \left[R_n(\rho) + \rho \frac{dR_n(\rho)}{d\rho} \right] \frac{\partial v(\rho, t)}{\partial \rho} d\rho.$$

Внеинтегральное слагаемое справа в последней формуле равно нулю в силу граничного условия. Далее, замечаем, что первый интеграл в формуле (***) и часть интеграла, стоящего в правой части последнего выражения, взаимно уничтожаются. Таким образом, вместо (***) будем иметь

$$\begin{aligned} \frac{1}{a^2} \frac{d^2 T}{dt^2} &= - \int_0^b \frac{\partial v(\rho, t)}{\partial \rho} \rho \frac{dR_n(\rho)}{d\rho} d\rho - \int_0^b \frac{v(\rho, t)}{\rho} R_n(\rho) d\rho + \\ &+ \sin(\omega t) \int_0^b \rho^2 J_1 \left(\frac{\mu_n \rho}{b} \right) d\rho. \end{aligned}$$

Первый интеграл справа еще раз преобразовываем по частям. Все эти преобразования проделаем в Maple. Выделяем множители подынтегральной функции в выражении _2 и переписываем интеграл в этом выражении как два отдельных интеграла

```
> op(1, (op(1, op(2, (op(2, _2))))));
op(2, (op(1, op(2, (op(2, _2))))));
```

$$R_n(\rho) + \rho \left(\frac{d}{d\rho} R_n(\rho) \right) \frac{\partial}{\partial \rho} v(\rho, t)$$

С учетом граничного условия будем иметь

```
> _2 := -Int(op(1, %%)*%, rho=0..b) - Int(op(2, %%)*%, rho=0..b);
```

$$_2 := - \int_0^b \left(\frac{\partial}{\partial \rho} v(\rho, t) \right) R_n(\rho) d\rho - \int_0^b \rho \left(\frac{d}{d\rho} R_n(\rho) \right) \left(\frac{\partial}{\partial \rho} v(\rho, t) \right) d\rho$$

Складываем теперь интегралы $_1$ и $_2$

```
> _1_2 := simplify(_2+_1);
```

$$_1_2 := - \int_0^b \rho \left(\frac{d}{d\rho} R_n(\rho) \right) \left(\frac{\partial}{\partial \rho} v(\rho, t) \right) d\rho$$

Берем полученный интеграл по частям

```
> _1_2 := student[intparts](_1_2, rho*diff(R[n](rho), rho));
```

$$_1_2 := -b \left(\frac{d}{db} R_n(b) \right) v(b, t) + \int_0^b \left(\left(\frac{d}{d\rho} R_n(\rho) \right) + \rho \left(\frac{d^2}{d\rho^2} R_n(\rho) \right) \right) v(\rho, t) d\rho$$

Внеинтегральное слагаемое исчезает в силу граничного условия

```
> _1_2 := op(2, _1_2);
```

$$_1_2 := \int_0^b \left(\left(\frac{d}{d\rho} R_n(\rho) \right) + \rho \left(\frac{d^2}{d\rho^2} R_n(\rho) \right) \right) v(\rho, t) d\rho$$

Таким образом, получаем

```
> ode2 := lhs(ode2) = _1_2+_3+_4;
```

$$\begin{aligned} \text{ode2} &:= \frac{d^2}{dt^2} T(t) = \int_0^b \left(\left(\frac{d}{d\rho} R_n(\rho) \right) + \rho \left(\frac{d^2}{d\rho^2} R_n(\rho) \right) \right) v(\rho, t) d\rho \\ &+ \int_0^b - \frac{v(\rho, t) R_n(\rho)}{\rho} d\rho + \int_0^b \rho^2 \sin(\omega t) \text{BesselJ} \left(1, \frac{\mu_n \rho}{b} \right) d\rho \end{aligned}$$

или, с учетом уравнения ode1 будем иметь для трансформанты Гринберга

```
> ode2:=lhs(ode2)+mu[n]^2/b^2*T(t)=
simplify(subs(BesselJ(1,mu[n])=
mu[n]*BesselJ(0,mu[n]),value(_4)));
```

$$ode2 := \frac{d^2}{dt^2} T(t) + \frac{\mu_n^2}{b^2} T(t) = \frac{b^3 \sin(\omega t) \text{BesselJ}(0, \mu_n)}{\mu_n}$$

Решаем теперь это уравнение с учетом начальных условий

```
> sol_ode2:=simplify(dsolve({ode2,T(0)=0,D(T)(0)=0},T(t)));
```

$$sol_ode2 := T(t) = - \frac{b^5 \text{BesselJ}(0, \mu_n) a \left(\sin\left(\frac{a \mu_n t}{b}\right) b \omega - \sin(\omega t) a \mu_n \right)}{\mu_n^2 (\mu_n^2 a^2 - \omega^2 b^2)}$$

```
> T:=unapply(rhs(%),n,t);
```

$$T := (n, t) \rightarrow - \frac{b^5 \text{BesselJ}(0, \mu_n) a \left(\sin\left(\frac{a \mu_n t}{b}\right) b \omega - \sin(\omega t) a \mu_n \right)}{\mu_n^2 (\mu_n^2 a^2 - \omega^2 b^2)}$$

Таким образом, мы получили трансформанту Гринберга в виде

$$T_n(t) = \frac{b^5 a J_0(\mu_n) \left[a \mu_n \sin(\omega t) - \omega b \sin\left(\frac{a \mu_n t}{b}\right) \right]}{\mu_n^2 (\mu_n^2 a^2 - \omega^2 b^2)}$$

Выпишем теперь решение задачи

```
> 'v(rho,t)'=Sum(T(n,t)*ef(rho,n)/norm2,n=1..infinity);
```

$$v(\rho, t) = \sum_{n=1}^{\infty} \left(- \frac{2 b^3 a \left(\sin\left(\frac{a \mu_n t}{b}\right) b \omega - \sin(\omega t) a \mu_n \right) \text{BesselJ}\left(1, \frac{\mu_n \rho}{b}\right)}{\text{BesselJ}(0, \mu_n) \mu_n^2 (\mu_n^2 a^2 - \omega^2 b^2) (-1 + \mu_n^2)} \right)$$

Итак, мы получили формальное решение для функции $v(\rho, t)$

$$v(\rho, t) = \sum_{n=1}^{\infty} \frac{2ab^3 \left[a \mu_n \sin(\omega t) - \omega b \sin\left(\frac{a \mu_n t}{b}\right) \right]}{\mu_n^2 (\mu_n^2 - 1) (\mu_n^2 a^2 - \omega^2 b^2) J_0(\mu_n)} J_1\left(\frac{\mu_n \rho}{b}\right).$$

Окончательно, решение задачи имеет вид

```
> v:=(rho,t)->Sum(T(n,t)*ef_n(rho,n),n=1..infinity):
```

```
'u(rho,t)'=u(rho,phi,t);
```

$$u(\rho, t) =$$

$$\cos(\phi) \left(\sum_{n=1}^{\infty} \left(- \frac{2 b^3 a \left(\sin\left(\frac{a \mu_n t}{b}\right) b \omega - \sin(\omega t) a \mu_n \right) \text{BesselJ}\left(1, \frac{\mu_n \rho}{b}\right)}{\text{BesselJ}(0, \mu_n) \mu_n^2 (\mu_n^2 a^2 - \omega^2 b^2) (-1 + \mu_n^2)} \right) \right)$$

то есть

$$u(\rho, t) = \cos(\phi) \sum_{n=1}^{\infty} \frac{2ab^3 \left[a\mu_n \sin(\omega t) - \omega b \sin\left(\frac{a\mu_n t}{b}\right) \right]}{\mu_n^2 (\mu_n^2 - 1) (\mu_n^2 a^2 - \omega^2 b^2) J_0(\mu_n)} J_1\left(\frac{\mu_n \rho}{b}\right),$$

где μ_n — последовательные положительные корни уравнения $J_1'(\mu) = 0$. Здесь мы предполагаем, что $\omega \neq \frac{\mu_n a}{b}$ для любого n .

Задача 5.6. Решить смешанную задачу

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + \frac{1}{x} \frac{\partial u}{\partial x} - \frac{9u}{x^2} + f(t) J_3(\mu_k x), \quad 0 < x < 1,$$

$$|u|_{x=0} < \infty, \quad u|_{x=1} = 0, \quad u|_{t=0} = 0, \quad \frac{\partial u}{\partial t} \Big|_{t=0} = 0,$$

где μ_k — положительный корень уравнения $J_3(\mu) = 0$, если:

1. $f(t) = e^{-t}$;
2. $f(t) = t - t^2$.

Решение. Определяем задачу в Maple:

```
> restart;
> eq:=diff(u(x,t),t,t)=diff(u(x,t),x,x)+
1/x*diff(u(x,t),x)-9/x^2*u(x,t)+f(t)*BesselJ(3,mu[k]*x);
eq := \frac{\partial^2}{\partial t^2} u(x,t) = \left( \frac{\partial^2}{\partial x^2} u(x,t) \right) + \frac{\frac{\partial}{\partial x} u(x,t)}{x} - \frac{9 u(x,t)}{x^2} + f(t) \text{BesselJ}(3, \mu_k x)
> ic1:=u(x,0)=0; ic2:=D[2](u)(x,0)=0;
ic1 := u(x,0) = 0
ic2 := D_2(u)(x,0) = 0
> bc:=u(1,t) = 0;
bc := u(1,t) = 0
```

Здесь мы имеем неоднородное уравнение при однородных дополнительных условиях. Решаем задачу методом Гринберга. Для этого сформулируем соответствующую задачу Штурма-Лиувилля.

Разделяем переменные в однородном уравнении:

```
> subs (f (t)=0, eq) ; pde_sol := pdsolve (%, HINT=X (x) *T (t) ) ;
> eq1 := op ([2, 1, 1], pde_sol) ; eq2 := op ([2, 1, 2], pde_sol) ;
```

$$\frac{\partial^2}{\partial t^2} u(x, t) = \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) + \frac{\partial}{\partial x} \frac{u(x, t)}{x} - \frac{9 u(x, t)}{x^2}$$

$pde_sol := (u(x, t) = X(x) T(t)) \&where$

$$\left[\left\{ \frac{d^2}{dx^2} X(x) = -X(x) _c_1 + \frac{-\left(\frac{d}{dx} X(x)\right) x + 9 X(x)}{x^2}, \frac{d^2}{dt^2} T(t) = -T(t) _c_1 \right\} \right]$$

$$eq1 := \frac{d^2}{dx^2} X(x) = -X(x) _c_1 + \frac{-\left(\frac{d}{dx} X(x)\right) x + 9 X(x)}{x^2}$$

$$eq2 := \frac{d^2}{dt^2} T(t) = -T(t) _c_1$$

Формулируем задачу Штурма-Лиувилля:

```
> eq1 := subs (_c [1]=lambda, eq1) ; s1 := X(1)=0;
```

$$eq1 := \frac{d^2}{dx^2} X(x) = -X(x) \lambda + \frac{-\left(\frac{d}{dx} X(x)\right) x + 9 X(x)}{x^2}$$

$$s1 := X(1) = 0$$

Таким образом, мы имеем следующую задачу Штурма-Лиувилля:

$$xX''(x) + X'(x) + \left(\lambda x - \frac{9}{x} \right) X(x) = 0, \quad (*)$$

$$|X|_{x=0} < \infty, \quad X|_{x=1} = 0.$$

Находим общее решение полученного уравнения Штурма-Лиувилля

```
> res1 := dsolve (eq1, X (x) ) ;
```

$$res1 := X(x) = _C1 \text{ BesselJ}(3, \sqrt{\lambda} x) + _C2 \text{ BesselY}(3, \sqrt{\lambda} x)$$

то есть мы получили

$$X(x) = C_1 J_3(\sqrt{\lambda} x) + C_2 Y_3(\sqrt{\lambda} x).$$

Из условия ограниченности решения в нуле находим $C_2 = 0$. А тогда получаем

$$X(x) = C_1 J_3(\sqrt{\lambda}x)$$

и из второго граничного условия находим ($C_1 \neq 0$)

$$J_3(\sqrt{\lambda}) = 0.$$

Пусть μ_n — последовательные положительные корни уравнения $J_3(\mu) = 0$; тогда собственные числа задачи образуют дискретный спектр значений $\lambda_n = \mu_n^2$, $n = 1, 2, 3, \dots$. Соответствующие собственные функции будут

$$X_n(x) = J_3(\mu_n x).$$

Определим эти функции в Maple:

```
> Xk:=unapply(
  subs(_C2=0, _C1=1, sqrt(lambda)=mu[k], rhs(%)), k, x);
  Xk := (k, x) -> BesselJ(3, mu_k x)
```

Согласно общей теории, найденные собственные функции должны быть ортогональны на отрезке $[0, 1]$ с весом $r(x) = x$. Проверим это

```
> simplify(BesselJ(3, mu[k]))=0:
> e1:=isolate(%, BesselJ(1, mu[k]));
e1n:=subs(k=n, e1); e1m:=subs(k=m, e1);
```

$$e1 := \text{BesselJ}(1, \mu_k) = -\frac{4 \text{BesselJ}(0, \mu_k) \mu_k}{-8 + \mu_k^2}$$

$$e1n := \text{BesselJ}(1, \mu_n) = -\frac{4 \text{BesselJ}(0, \mu_n) \mu_n}{-8 + \mu_n^2}$$

$$e1m := \text{BesselJ}(1, \mu_m) = -\frac{4 \text{BesselJ}(0, \mu_m) \mu_m}{-8 + \mu_m^2}$$

```
> Int(x*Xk(n, x)*Xk(m, x), x=0..1);
```

$$\int_0^1 x \text{BesselJ}(3, \mu_n x) \text{BesselJ}(3, \mu_m x) dx$$

```
> simplify(subs({e1m, e1n}, value(%)));
```

0

Итак, мы получили формулу

$$\int_0^1 x J_3(\mu_m x) J_3(\mu_n x) dx = 0, \quad m \neq n,$$

которая и доказывает ортогональность функций.

Найдем теперь квадрат нормы собственных функций, то есть, вычислим интеграл

$$\|X_n\|^2 = \int_0^1 x J_3^2(\mu_n x) dx.$$

```
> norm2 := Int(x * Xk(k, x)^2, x=0..1);
simplify(subs({e1}, value(%)));
```

$$\begin{aligned} norm2 &:= \int_0^1 x \text{BesselJ}(3, \mu_k x)^2 dx \\ &= \frac{1}{2} \frac{\text{BesselJ}(0, \mu_k)^2 \mu_k^4}{(-8 + \mu_k^2)^2} \end{aligned}$$

Итак, мы получили

$$\|X_n\|^2 = \int_0^1 x J_3^2(\mu_n x) dx = \frac{1}{2} \frac{\mu_n^4}{(\mu_n^2 - 8)^2} J_0^2(\mu_n).$$

Следовательно, нормированные собственные функции задачи будут

$$\tilde{X}_n(x) = \frac{2(\mu_n^2 - 8)^2}{\mu_n^4 J_0^2(\mu_n)} J_3(\mu_n x)$$

то есть

```
> X_norm := unapply(Xk(k, x) / norm2, k, x);
```

$$X_norm := (k, x) \rightarrow \frac{2 \text{BesselJ}(3, \mu_k x) (-8 + \mu_k^2)^2}{\text{BesselJ}(0, \mu_k)^2 \mu_k^4}$$

Решение исходной задачи ищем в виде разложения

$$u(x, t) = \sum_{n=1}^{\infty} \frac{T_n(t) X_n(x)}{\|X_n\|^2}, \quad T_n(t) = \int_0^1 u(x, t) x X_n(x) dx. \quad (**)$$

Получим уравнение для трансформанты Гринберга $T_n(t)$. Умножим исходное уравнение на $x X_n(x)$ и интегрируем на отрезке $[0, 1]$:

```
> ode2 := Int(lhs(eq) * x * X[n](x), x=0..1) =
```

```

Int (op (1, rhs (eq)) *x* 'X[n] (x) ', x=0..1) +
Int (op (2, rhs (eq)) *x* 'X[n] (x) ', x=0..1) +
Int (op (3, rhs (eq)) *x* 'X[n] (x) ', x=0..1) +
Int (op (4, rhs (eq)) *x* 'X[n] (x) ', x=0..1) ;

```

$$\begin{aligned}
ode2 := & \int_0^1 \left(\frac{\partial^2}{\partial t^2} u(x, t) \right) x X_n(x) dx = \int_0^1 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) x X_n(x) dx \\
& + \int_0^1 \left(\frac{\partial}{\partial x} u(x, t) \right) X_n(x) dx + \int_0^1 -\frac{9 u(x, t) X_n(x)}{x} dx \\
& + \int_0^1 f(t) \text{BesselJ}(3, \mu_k x) x X_n(x) dx
\end{aligned}$$

```
> ode2 := diff (T[n] (t) , t$2) = rhs (ode2) ;
```

$$\begin{aligned}
ode2 := & \frac{d^2}{dt^2} T_n(t) = \int_0^1 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) x X_n(x) dx + \int_0^1 \left(\frac{\partial}{\partial x} u(x, t) \right) X_n(x) dx \\
& + \int_0^1 -\frac{9 u(x, t) X_n(x)}{x} dx + \int_0^1 f(t) \text{BesselJ}(3, \mu_k x) x X_n(x) dx
\end{aligned}$$

Преобразовываем первый интеграл справа по частям. Для этого выделим в отдельные выражения интегралы в правой части

```

_1 := op (1, rhs (ode2)) ; _2 := op (2, rhs (ode2)) ;
_3 := op (3, rhs (ode2)) ; _4 := op (4, rhs (ode2)) ;

```

$$_1 := \int_0^1 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) x X_n(x) dx$$

$$_2 := \int_0^1 \left(\frac{\partial}{\partial x} u(x, t) \right) X_n(x) dx$$

$$_3 := \int_0^1 -\frac{9 u(x, t) X_n(x)}{x} dx$$

$$_4 := \int_0^1 f(t) \text{BesselJ}(3, \mu_k x) x X_n(x) dx$$

Берем первый интеграл по частям

```
> _1:=student[intparts](_1,x*X[n](x));
```

$$_I := X_n(1) D_1(u)(1, t) - \int_0^1 \left(X_n(x) + x \left(\frac{d}{dx} X_n(x) \right) \right) \left(\frac{\partial}{\partial x} u(x, t) \right) dx$$

Полученный интеграл представим как два отдельных интеграла и скомбинируем его с интегралом _2

```
> op(1, (op(1, op(2, (op(2, _1))))));
```

```
op(2, (op(1, op(2, (op(2, _1))))));
```

$$X_n(x) + x \left(\frac{d}{dx} X_n(x) \right) \frac{\partial}{\partial x} u(x, t)$$

```
> _1:=op(1, _1) - Int(op(1, %%)*%, x=0..1) -
```

```
Int(op(2, %%)*%, x=0..1);
```

$$_I := X_n(1) D_1(u)(1, t) - \int_0^1 \left(\frac{\partial}{\partial x} u(x, t) \right) X_n(x) dx - \int_0^1 x \left(\frac{d}{dx} X_n(x) \right) \left(\frac{\partial}{\partial x} u(x, t) \right) dx$$

```
> _1_2:=simplify(_1+_2);
```

$$_I_2 := X_n(1) D_1(u)(1, t) - \int_0^1 x D(X_n)(x) D_1(u)(x, t) dx$$

Учтем граничные условия и еще раз преобразуем по частям

```
> _1_2:=op(2, _1_2);
```

$$_I_2 := - \int_0^1 x D(X_n)(x) D_1(u)(x, t) dx$$

```
> _1_2:=student[intparts](_1_2,x*diff(X[n](x),x));
```

$$_I_2 := -D(X_n)(1) u(1, t) + \int_0^1 \left(\left(\frac{d}{dx} X_n(x) \right) + x \left(\frac{d^2}{dx^2} X_n(x) \right) \right) u(x, t) dx$$

или, с учетом граничных условий

```
> _1_2:=op(2, _1_2);
```

$$_1_2 := \int_0^1 \left(\left(\frac{d}{dx} X_n(x) \right) + x \left(\frac{d^2}{dx^2} X_n(x) \right) \right) u(x, t) dx$$

Таким образом, мы получили

> ode2 := lhs(ode2) = _1_2 + op(3, rhs(ode2)) + op(4, rhs(ode2));

$$\begin{aligned} \text{ode2} := \frac{d^2}{dt^2} T_n(t) &= \int_0^1 \left(\left(\frac{d}{dx} X_n(x) \right) + x \left(\frac{d^2}{dx^2} X_n(x) \right) \right) u(x, t) dx \\ &+ \int_0^1 -\frac{9 u(x, t) X_n(x)}{x} dx + \int_0^1 f(t) \text{BesselJ}(3, \mu_k x) x X_n(x) dx \end{aligned}$$

Или, с учетом уравнения (*)

> eq1;

$$\frac{d^2}{dx^2} X(x) = -X(x) \lambda + \frac{-\left(\frac{d}{dx} X(x)\right) x + 9 X(x)}{x^2}$$

получаем окончательное уравнение для трансформанты Гринберга

> ode2 := lhs(ode2) + mu[n]^2 * T[n](t) = op(3, rhs(ode2));

$$\text{ode2} := \left(\frac{d^2}{dt^2} T_n(t) \right) + \mu_n^2 T_n(t) = \int_0^1 f(t) \text{BesselJ}(3, \mu_k x) x X_n(x) dx$$

Вычисляем интеграл справа

> subs(X[n](x) = Xk(n, x), rhs(ode2));

$$\int_0^1 f(t) \text{BesselJ}(3, \mu_k x) x \text{BesselJ}(3, \mu_n x) dx$$

> simplify(subs({e1, e1n}, value(%)));

0

То есть, мы получили

$$\int_0^1 f(t) J_3(\mu_k x) x X_n(x) dx = 0$$

если $n \neq k$.

Найдем теперь значение этого интеграла при $n = k$

> subs(X[n](x) = Xk(k, x), rhs(ode2));

simplify(subs({e1}, value(%)));

$$\int_0^1 f(t) \text{BesselJ}(3, \mu_k x)^2 x dx$$

$$\frac{1}{2} \frac{f(t) \text{BesselJ}(0, \mu_k)^2 \mu_k^4}{(-8 + \mu_k^2)^2}$$

Итак, окончательно уравнение для трансформанты Гринберга имеет вид

> ode2 := subs({n=k}, lhs(ode2)) = %;

$$\text{ode2} := \left(\frac{d^2}{dt^2} T_k(t) \right) + \mu_k^2 T_k(t) = \frac{1}{2} \frac{f(t) \text{BesselJ}(0, \mu_k)^2 \mu_k^4}{(-8 + \mu_k^2)^2}$$

то есть

$$\frac{d^2 T_k}{dt^2} + \mu_k^2 T_k = \frac{1}{2} \frac{\mu_n^4}{(\mu_n^2 - 8)^2} f(t).$$

Рассмотрим два случая задания функции $f(t)$: первый вариант

> f := t -> exp(-t);

$$f := t \rightarrow e^{(-t)}$$

Тогда уравнение для трансформанты Гринберга принимает вид

> ode2_1 := subs({n=k}, ode2);

$$\text{ode2}_1 := \left(\frac{d^2}{dt^2} T_k(t) \right) + \mu_k^2 T_k(t) = \frac{1}{2} \frac{e^{(-t)} \text{BesselJ}(0, \mu_k)^2 \mu_k^4}{(-8 + \mu_k^2)^2}$$

Решаем это уравнение с учетом нулевых начальных условий

> sol_ode2_1 :=

simplify(dsolve({ode2_1, T[k](0)=0, D(T[k])(0)=0}, T[k](t)));

$$\text{sol_ode2}_1 := T_k(t) = \frac{1}{2} \frac{\text{BesselJ}(0, \mu_k)^2 \mu_k^3 (\sin(\mu_k t) - \cos(\mu_k t)) \mu_k + e^{(-t)} \mu_k}{48 \mu_k^2 - 15 \mu_k^4 + \mu_k^6 + 64}$$

> sol_ode2_1 := factor(sol_ode2_1);

$$\text{sol_ode2}_1 := T_k(t) = \frac{1}{2} \frac{\text{BesselJ}(0, \mu_k)^2 \mu_k^3 (\sin(\mu_k t) - \cos(\mu_k t)) \mu_k + e^{(-t)} \mu_k}{(\mu_k^2 + 1) (-8 + \mu_k^2)^2}$$

Таким образом, в сумме (**) сохраняется только одно слагаемое, соответствующее $n = k$ и мы получаем решение исходной задачи в виде

> sol_1:=factor(rhs(sol_ode2_1)*Xk(k,x)/norm2);

$$sol_1 := \frac{(\sin(\mu_k t) - \cos(\mu_k t) \mu_k + e^{(-t)} \mu_k) \text{BesselJ}(3, \mu_k x)}{\mu_k (\mu_k^2 + 1)}$$

Рассмотрим теперь второй вариант — функция задана так

> f:=t->t-t^2;

$$f := t \rightarrow t - t^2$$

> ode2_2:=subs({n=k},ode2);

$$ode2_2 := \left(\frac{d^2}{dt^2} T_k(t) \right) + \mu_k^2 T_k(t) = \frac{1}{2} \frac{(t - t^2) \text{BesselJ}(0, \mu_k)^2 \mu_k^4}{(-8 + \mu_k^2)^2}$$

> sol_ode2_2:=

factor(dsolve({ode2_2, T[k](0)=0, D(T[k])(0)=0}, T[k](t)));

sol_ode2_2 :=

$$T_k(t) = -\frac{1}{2} \frac{(-t \mu_k^2 + t^2 \mu_k^2 + \sin(\mu_k t) \mu_k + 2 \cos(\mu_k t) - 2) \text{BesselJ}(0, \mu_k)^2}{(-8 + \mu_k^2)^2}$$

Таким образом, и в этом случае в сумме (**) сохраняется только одно слагаемое, соответствующее $n = k$ и мы получаем решение исходной задачи в виде

> sol_2:=factor(rhs(sol_ode2_2)*Xk(k,x)/norm2);

$$sol_2 := -\frac{(-t \mu_k^2 + t^2 \mu_k^2 + \sin(\mu_k t) \mu_k + 2 \cos(\mu_k t) - 2) \text{BesselJ}(3, \mu_k x)}{\mu_k^4}$$

Подведем итог: если $f(t) = e^{-t}$, то решение задачи имеет вид

> 'u(x,t) '=sol_1;

$$u(x, t) = \frac{(\sin(\mu_k t) - \cos(\mu_k t) \mu_k + e^{(-t)} \mu_k) \text{BesselJ}(3, \mu_k x)}{\mu_k (\mu_k^2 + 1)}$$

то есть

$$u(x, t) = \frac{(\sin(\mu_k t) - \mu_k \cos(\mu_k t) + \mu_k e^{-t})}{\mu_k (\mu_k^2 + 1)} J_3(\mu_k x).$$

Если $f(t) = t - t^2$, то решение имеет вид

> 'u(x,t) '=sol_2;

$$u(x, t) = - \frac{(-t \mu_k^2 + t^2 \mu_k^2 + \sin(\mu_k t) \mu_k + 2 \cos(\mu_k t) - 2) \text{BesselJ}(3, \mu_k x)}{\mu_k^4}$$

то есть

$$u(x, t) = \frac{(\mu_k \sin(\mu_k t) + 2 \cos(\mu_k t) - 2 - \mu_k^2 t + \mu_k^2 t^2)}{\mu_k^4} J_3(\mu_k x).$$

Выполним проверку полученного решения

> eq;

$$\frac{\partial^2}{\partial t^2} u(x, t) = \left(\frac{\partial^2}{\partial x^2} u(x, t) \right) + \frac{\partial}{\partial x} \frac{u(x, t)}{x} - \frac{9 u(x, t)}{x^2} + f(t) \text{BesselJ}(3, \mu_k x)$$

> lprint(`Первый вариант:`);

u:=unapply(sol_1, x, t);

simplify(subs({f(t)=exp(-t)}, lhs(eq)-rhs(eq)));

simplify(ic1);

simplify(ic2);

simplify(bc);

`Первый вариант:`

$$u := (x, t) \rightarrow \frac{(\sin(\mu_k t) - \cos(\mu_k t) \mu_k + e^{(-t)} \mu_k) \text{BesselJ}(3, \mu_k x)}{\mu_k (\mu_k^2 + 1)}$$

0

0 = 0

0 = 0

- (\sin(\mu_k t) - \cos(\mu_k t) \mu_k + e^{(-t)} \mu_k)

(-8 BesselJ(1, \mu_k) + 4 BesselJ(0, \mu_k) \mu_k + BesselJ(1, \mu_k) \mu_k^2) / ((\mu_k^2 + 1)

\mu_k^3) = 0

> simplify(subs(e1, %));

0 = 0

> lprint(`Второй вариант:`);

u:=unapply(sol_1, x, t);

simplify(subs({f(t)=exp(-t)}, lhs(eq)-rhs(eq)));

simplify(ic1);

simplify(ic2);

simplify(bc);

`Второй вариант:`

$$u := (x, t) \rightarrow \frac{(\sin(\mu_k t) - \cos(\mu_k t) \mu_k + e^{(-t) \mu_k}) \text{BesselJ}(3, \mu_k x)}{\mu_k (\mu_k^2 + 1)}$$

0

0 = 0

0 = 0

$$-(\sin(\mu_k t) - \cos(\mu_k t) \mu_k + e^{(-t) \mu_k})$$

$$(-8 \text{BesselJ}(1, \mu_k) + 4 \text{BesselJ}(0, \mu_k) \mu_k + \text{BesselJ}(1, \mu_k) \mu_k^2) / ((\mu_k^2 + 1)$$

$$\mu_k^3) = 0$$

> simplify(subs(e1, %));

0 = 0

Видим, что в обоих вариантах и уравнение и все дополнительные условия (начальные и граничные условия) выполняются.

Упражнения

1. Найти распределение температуры в пластине толщиной a , грань $x = 0$ которой излучает тепло в окружающую среду по закону Ньютона (температура окружающей среды равна нулю), а другая грань $x = a$ поддерживается при постоянной температуре T_0 , равной начальной температуре пластины.

$$\text{Ответ. } u(x, \tau) = T_0 \left\{ \frac{1 + hx}{1 + ha} - 2 \sum_{n=1}^{\infty} \frac{\cos(\mu_n) e^{-\frac{\mu_n^2 \tau}{a^2}} \sin\left(\frac{\mu_n (a-x)}{a}\right)}{\mu_n \left(1 - \frac{\sin(2\mu_n)}{2\mu_n}\right)} \right\}, \quad \text{где } \mu_n \text{ —}$$

последовательные положительные корни уравнения $\text{tg } \mu = -\frac{\mu}{ah}$; h — коэффициент

теплообмена в законе Ньютона; $\tau = \frac{kt}{c\rho} = v^2 t$ — приведенное время.

2. Решить краевую задачу в шаре радиуса b :

$$\frac{1}{a^2} \frac{\partial^2 u}{\partial t^2} = \Delta u, \quad u|_{t=0} = 0, \quad \frac{\partial u}{\partial t} \Big|_{t=0} = 0, \quad \frac{\partial u}{\partial r} \Big|_{r=b} = A \cos \theta \sin \omega t.$$

Ответ. $u(r, \theta, t) = Ar \cos(\theta) \sin(\omega t) +$

$$+ \cos(\theta) \sum_{n=1}^{\infty} \frac{A\omega b C_n}{\mu_n^2 a^2 - \omega^2 b^2} \left[\mu_n a \sin\left(\frac{\mu_n a t}{b}\right) - \omega b \sin(\omega t) \right] R_n(r),$$

где μ_n — последовательные положительные корни уравнения $\operatorname{tg} \mu = -\frac{2\mu}{\mu^2 - 2}$;

$$C_n = \frac{2b^2 [\sin(\mu_n) - \mu_n \cos(\mu_n)]}{\mu_n [\mu_n^3 - 2\mu_n + 2 \cos(\mu_n) \sin(\mu_n)]}, \quad R_n(r) = -\frac{\left[\cos\left(\frac{\mu_n r}{b}\right) \mu_n r - \sin\left(\frac{\mu_n r}{b}\right) b \right]}{r^2}.$$

3. Решить смешанную задачу для неоднородного уравнения гиперболического типа

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + 2b, \quad b = \text{const}, \quad 0 < x < l, \quad t > 0;$$

$$u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}|_{t=0} = 0; \quad u|_{x=0} = 0, \quad u|_{x=l} = 0.$$

Ответ. $u(x, t) = bx(l-x) + \frac{4l^2 b}{\pi^2} \sum_{k=1}^{\infty} \frac{(-1)^k \cos\left(\frac{k\pi t}{l}\right) \sin\left(\frac{k\pi x}{l}\right)}{k^2}$. Указание. Решение

можно искать в виде $u(x, t) = v(x) + w(x, t)$, где функция $v(x) = bx(l-x)$ удовлетворяет неоднородному уравнению и нулевым граничным условиям, а функция $w(x, t)$ удовлетворяет однородному уравнению, нулевым граничным условиям и следующим начальным условиям: $w(x, 0) = -bx(l-x)$, $\frac{\partial w}{\partial t}|_{t=0} = 0$.

4. Решить смешанную задачу для неоднородного уравнения гиперболического типа

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + \cos(t), \quad 0 < x < \pi, \quad t > 0;$$

$$u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}|_{t=0} = 0; \quad u|_{x=0} = 0, \quad u|_{x=\pi} = 0.$$

Ответ. $u(x, t) = \frac{2}{\pi} t \sin t \sin x + \sum_{k=2}^{\infty} \frac{4 [\cos t - \cos(kt)] \sin(kx)}{\pi k (1 - k^2)}$.

5. Решить задачу о вынужденных поперечных колебаниях струны, закрепленной на одном конце $x=0$ и подверженной на другом конце $x=l$ действию возмущающей силы, которая вызывает смещение, равное $A \sin(\omega t)$, где $\omega \neq k\pi v/l$, $k = 1, 2, 3, \dots$. Начальные условия — нулевые.

$$\text{Ответ. } u(x, t) = A \frac{\sin \frac{\omega x}{v}}{\sin \frac{\omega l}{v}} \sin(\omega t) + \frac{2A\omega v}{l} \sum_{k=1}^{\infty} \frac{(-1)^{k-1} \sin\left(\frac{k\pi vt}{l}\right) \sin\left(\frac{k\pi x}{l}\right)}{\omega^2 - \frac{k^2 \pi^2 v^2}{l^2}}.$$

Указание. Решение можно искать в виде $u(x, t) = v(x, t) + w(x, t)$, где функция $v(x, t) = X(x) \sin(\omega t)$. Эту функцию подобрать так, чтобы она удовлетворяла уравнению и заданным граничным условиям.

6. Решить смешанную задачу для уравнения гиперболического типа

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}, \quad 0 < x < l, \quad t > 0;$$

$$u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}|_{t=0} = 0; \quad u|_{x=0} = 0, \quad u|_{x=l} = t.$$

$$\text{Ответ. } u(x, t) = \frac{xt}{l} + \frac{2l}{\pi^2} \sum_{k=1}^{\infty} \frac{(-1)^k \sin\left(\frac{k\pi t}{l}\right) \sin\left(\frac{k\pi x}{l}\right)}{k^2}.$$

7. Решить смешанную задачу для уравнения гиперболического типа

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}, \quad 0 < x < 1, \quad t > 0;$$

$$u|_{t=0} = x + 1, \quad \frac{\partial u}{\partial t}|_{t=0} = 0; \quad u|_{x=0} = t + 1, \quad u|_{x=1} = t^3 + 2.$$

$$\text{Ответ. } u(x, t) = t + 1 + x(t^3 - t + 1) +$$

$$+ \sum_{k=1}^{\infty} \left\{ \frac{2}{(\pi k)^2} \left[\frac{6(-1)^{k+1}}{(\pi k)^2} - 1 \right] \sin(\pi kt) + \frac{(-1)^k 12t}{(\pi k)^3} \right\} \sin(\pi kx).$$

8. Решить смешанную задачу для уравнения гиперболического типа

$$\frac{\partial^2 u}{\partial t^2} = a^2 \frac{\partial^2 u}{\partial x^2}, \quad 0 < x < l, \quad t > 0;$$

$$u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}|_{t=0} = 0; \quad u|_{x=0} = 0, \quad \frac{\partial u}{\partial x}|_{x=l} = \frac{Q}{ES} = \text{const}.$$

$$\text{Ответ. } u(x, t) = \frac{Qx}{ES} - \frac{8Ql}{\pi^2 ES} \sum_{k=0}^{\infty} \frac{(-1)^k \sin\left(\frac{(2k+1)\pi x}{2l}\right) \cos\left(\frac{(2k+1)a\pi t}{2l}\right)}{(2k+1)^2}.$$

9. Решить смешанную задачу для уравнения гиперболического типа

$$\frac{\partial^2 u}{\partial t^2} = a^2 \frac{\partial^2 u}{\partial x^2}, \quad 0 < x < l, \quad t > 0;$$

$$u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}|_{t=0} = 0; \quad u|_{x=0} = 0, \quad \frac{\partial u}{\partial x}|_{x=l} = \frac{A \sin \omega t}{ES}.$$

Ответ. $u(x, t) = \frac{Aa}{ES\omega} \frac{\sin\left(\frac{\omega x}{a}\right) \sin(\omega t)}{\cos\left(\frac{\omega l}{a}\right)} -$

$$- \frac{4A\omega}{ES\pi} \sum_{k=0}^{\infty} \frac{(-1)^k \sin\left(\frac{(2k+1)\pi x}{2l}\right) \sin\left(\frac{(2k+1)\pi t}{2l}\right)}{(2k+1) \left[\omega^2 - \frac{(2k+1)^2 \pi^2 a^2}{4l^2} \right]}.$$

Предполагается, что $\omega \neq \frac{(2k+1)\pi a}{2l}$, $k = 0, 1, \dots$

10. Найти стационарное распределение температуры в бруске, сечение которого имеет форму криволинейного прямоугольника ($a \leq r \leq b$, $0 \leq \varphi \leq \alpha$) при следующих условиях

$$u|_{r=a} = 0; \quad u|_{r=b} = T_0; \quad u|_{\varphi=0} = 0; \quad u|_{\varphi=\alpha} = T_0.$$

Ответ. $u(r, \varphi) = T_0 \left\{ \frac{\ln\left(\frac{r}{a}\right)}{\ln\left(\frac{b}{a}\right)} + \right.$

$$\left. + \frac{2}{\pi} \sum_{n=1}^{\infty} \frac{(-1)^n \operatorname{sh}\left(\frac{n\pi(\alpha - \varphi)}{\ln\left(\frac{b}{a}\right)}\right) + \operatorname{sh}\left(\frac{n\pi\varphi}{\ln\left(\frac{b}{a}\right)}\right)}{n \operatorname{sh}\left(\frac{n\pi\alpha}{\ln\left(\frac{b}{a}\right)}\right)} \sin\left(n\pi \frac{\ln\left(\frac{r}{a}\right)}{\ln\left(\frac{b}{a}\right)}\right) \right\}.$$

Другая форма решения:

$$u(r, \varphi) = T_0 \left\{ \frac{\varphi}{\alpha} + \right.$$

$$+ \frac{2}{\pi} \sum_{n=1}^{\infty} \left[\frac{1 - (-1)^n \left(\frac{a}{b}\right)^{\frac{n\pi}{\alpha}} \left(\frac{r}{b}\right)^{\frac{n\pi}{\alpha}} - \left[1 - (-1)^n \left(\frac{b}{a}\right)^{\frac{n\pi}{\alpha}} \left(\frac{a}{r}\right)^{\frac{n\pi}{\alpha}} \right]}{n \left[\left(\frac{b}{a}\right)^{\frac{n\pi}{\alpha}} - \left(\frac{a}{b}\right)^{\frac{n\pi}{\alpha}} \right]} \sin\left(\frac{n\pi\varphi}{\alpha}\right) \right].$$

11. Изучить вынужденные поперечные колебания струны, закрепленной на концах $x = 0$, $x = l$ и подверженной действию внешней возмущающей силы $F_0 \sin(\omega t)$, $\omega \neq k\pi v/l$, $k = 1, 2, 3, \dots$, сосредоточенной в точке $x = c$, $c \in (0, l)$ струны.

Ответ. $u(x, t) = \frac{2Av}{\pi\omega T} \sum_{k=1}^{\infty} \frac{\sin\left(\frac{k\pi vt}{l}\right) - \frac{k\pi v}{\omega l} \sin(\omega t) \sin\left(\frac{k\pi c}{l}\right) \sin\left(\frac{k\pi x}{l}\right)}{\left[1 - \frac{k^2\pi^2 v^2}{\omega^2 l^2}\right] k}$, где $v = \sqrt{\frac{T}{\rho}}$,

T — натяжение струны, ρ — линейная плотность струны.

12. Решить краевую задачу

$$\frac{\partial^2 u}{\partial t^2} = a^2 \frac{\partial^2 u}{\partial x^2}, \quad 0 < x < l, \quad t > 0,$$

$$u|_{x=0} = A, \quad u|_{x=l} = B, \quad u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}\bigg|_{t=0} = 0.$$

Ответ. $u(x, t) = A + \frac{B-A}{l}x + \sum_{n=1}^{\infty} \frac{2[A + (-1)^{n+1}B]}{n\pi} \sin\left(\frac{n\pi x}{l}\right) \cos\left(\frac{n\pi at}{l}\right).$

13. Решить краевую задачу

$$\frac{\partial^2 u}{\partial t^2} = a^2 \frac{\partial^2 u}{\partial x^2}, \quad 0 < x < l, \quad t > 0,$$

$$\frac{\partial u}{\partial x}\bigg|_{x=0} = A, \quad \frac{\partial u}{\partial x}\bigg|_{x=l} = B, \quad u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}\bigg|_{t=0} = 0.$$

Ответ. $u(x, t) = \frac{Bx^2}{2l} + Ax + \frac{Ba^2 t^2}{2l} - l^2 \left(\frac{A}{B} + \frac{1}{3}\right) +$
 $+ \frac{4l^2}{\pi^2} \sum_{n=1}^{\infty} \left\{ \frac{A}{B} [1 + (-1)^{n+1}] + (-1)^{n+1} \right\} \cos\left(\frac{n\pi x}{l}\right) \cos\left(\frac{n\pi vt}{l}\right).$

14. Решить краевую задачу

$$\frac{1}{a^2} \frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + A, \quad 0 < x < l, \quad t > 0$$

$$\left. \frac{\partial u}{\partial x} \right|_{x=0} = 0, \quad \left. \frac{\partial u}{\partial x} \right|_{x=l} = 0, \quad u|_{t=0} = 0, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = 0.$$

Ответ. $u = \frac{Aa^2}{2} t^2.$

15. Решить уравнение

$$\frac{1}{a^2} \frac{\partial^2 u}{\partial t^2} = \Delta u + A(1 - \rho^2) \sin(\omega t)$$

в круге радиуса $R = 1$ при краевом условии $u(\rho, t)|_{\rho=1} = 0$ и при нулевых начальных условиях $u(\rho, t)|_{t=0} = 0, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = 0.$

Ответ. $u(\rho, t) = \sum_{n=1}^{\infty} \frac{8A}{\mu_n^3 J_1(\mu_n)(\mu_n^2 - \omega^2)} \left[\sin(\omega t) - \frac{\omega \sin(\mu_n a t)}{a \mu_n} \right] J_0(\mu_n \rho), \quad \omega \neq \mu_n.$

16. Решить краевую задачу

$$\frac{1}{a^2} \frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}, \quad 0 < x < l, \quad t > 0$$

$$u|_{x=0} = A \cos \omega t, \quad \left. \frac{\partial u}{\partial x} \right|_{x=l} = 0, \quad u|_{t=0} = 0, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = 0.$$

17. Решить краевую задачу в полукруге радиуса b

$$\frac{1}{a^2} \frac{\partial^2 u}{\partial t^2} = \Delta u, \quad 0 \leq \rho \leq b, \quad 0 \leq \varphi \leq \pi;$$

$$u(\rho, \varphi, t)|_{\rho=b} = b \sin(\varphi), \quad u(\rho, 0, t) = u(\rho, \pi, t) = 0, \quad u|_{t=0} = 0, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = 0.$$

Ответ. $u(\rho, \varphi, t) = \rho \sin(\varphi) + \sin(\varphi) \sum_{n=1}^{\infty} \frac{2b}{\mu_n J_0(\mu_n)} J_1\left(\frac{\mu_n \rho}{b}\right) \cos\left(\frac{a \mu_n t}{b}\right),$ где μ_n — последовательные положительные корни уравнения $J_1(\mu) = 0.$

18. Решить краевую задачу в круге радиуса b

$$\frac{1}{a^2} \frac{\partial^2 u}{\partial t^2} = \Delta u + A \sin(\omega t),$$

$$u(\rho, t)|_{\rho=b} = 0, \quad u|_{t=0} = 0, \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = 0.$$

Ответ. $u(\rho, t) = \sum_{n=1}^{\infty} \left[B_n \sin\left(\frac{a\mu_n t}{b}\right) + C_n \sin(\omega t) \right] J_0\left(\frac{\mu_n \rho}{b}\right)$, где μ_n —
 последовательные положительные корни уравнения $J_0(\mu) = 0$,
 $C_n = \frac{2A}{\mu_n J_1(\mu_n) \left[\frac{\mu_n^2}{b^2} - \frac{\omega^2}{a^2} \right]}$, $B_n = -\frac{\omega b \mu_n}{a} C_n$, $\omega \neq \frac{a\mu_n}{b}$.

19. Решить краевую задачу в цилиндре радиуса b , высоты h

$$\frac{1}{a^2} \frac{\partial^2 u}{\partial t^2} = \Delta u, \quad u|_{t=0} = 0, \quad \frac{\partial u}{\partial t} \Big|_{t=0} = 0,$$

$$u(\rho, \varphi, z, t)|_{z=0} = A, \quad u(\rho, \varphi, z, t)|_{z=h} = B, \quad u(\rho, \varphi, z, t)|_{\rho=b} = 0.$$

Указание. $u = w + u_1$, $\Delta u_1 = 0$, $u_1|_{z=0} = A$, $u_1|_{z=h} = B$, $u_1|_{\rho=b} = 0$,

$$u_1 = \sum_{n=1}^{\infty} \left[A_n \operatorname{ch}\left(\frac{z\mu_n}{b}\right) + B_n \operatorname{sh}\left(\frac{z\mu_n}{b}\right) \right] J_0\left(\frac{\mu_n \rho}{b}\right), \quad A_n, B_n \text{ определяются из разложений}$$

$$A = \sum_{n=1}^{\infty} A_n J_0\left(\frac{\mu_n \rho}{b}\right), \quad B = \sum_{n=1}^{\infty} \left[A_n \operatorname{ch}\left(\frac{h\mu_n}{b}\right) + B_n \operatorname{sh}\left(\frac{h\mu_n}{b}\right) \right] J_0\left(\frac{\mu_n \rho}{b}\right),$$

$$w = \sum_{n,m=1}^{\infty} A_{n,m} \sin\left(\frac{m\pi z}{h}\right) \cos \left[at \sqrt{\left(\frac{m\pi}{h}\right)^2 + \left(\frac{\mu_n}{b}\right)^2} \right] J_0\left(\frac{\mu_n \rho}{b}\right), \quad A_{n,m} \text{ определяются из}$$

условия $w(\rho, z, 0) = -u_1(\rho, z)$.

20. Решить краевую задачу в цилиндре радиуса b , высоты h

$$\frac{1}{a^2} \frac{\partial u}{\partial t} = \Delta u, \quad u|_{t=0} = 0,$$

$$u|_{z=0} = 0, \quad u|_{z=h} = 0, \quad u|_{\rho=b} = kt \sin\left(\frac{\pi z}{h}\right).$$

Указание. $u = w + u_1$, $u_1 = kt \sin\left(\frac{\pi z}{h}\right)$;

$$\frac{1}{a^2} \frac{\partial w}{\partial t} = \Delta w + f, \quad \text{где } f = -k \sin\left(\frac{\pi z}{h}\right) \left[\frac{\pi^2}{h^2} t + \frac{1}{a^2} \right].$$

21. Решить уравнение Пуассона в параллелепипеде $0 \leq x \leq a$, $0 \leq y \leq b$, $0 \leq z \leq c$

$$\Delta u = A,$$

$$u|_{x=0} = 0, \quad u|_{x=a} = B, \quad u|_{y=0} = 0, \quad u|_{y=b} = 0, \quad u|_{z=0} = 0, \quad u|_{z=c} = 0.$$

Указание. $u = w + u_1$, $\Delta u_1 = 0$, $u_1|_{x=a} = B$ и $u_1 = 0$ на остальной части границы;

$$u_1 = \sum_{n,m=1}^{\infty} A_{n,m} \sin\left(\frac{(2n-1)\pi y}{b}\right) \sin\left(\frac{(2m-1)\pi z}{c}\right) \operatorname{sh}(\pi x) \sqrt{\left(\frac{2m-1}{c}\right)^2 + \left(\frac{2n-1}{b}\right)^2},$$

$$w = \sum_{n,m,k=1}^{\infty} B_{n,m,k} \sin\left(\frac{(2n-1)\pi y}{b}\right) \sin\left(\frac{(2m-1)\pi z}{c}\right) \sin\left(\frac{(2k-1)\pi x}{a}\right),$$

$$A_{n,m} = 4B \left[\pi^2 (2n-1)(2m-1) \operatorname{sh}(\pi a) \sqrt{\left(\frac{2m-1}{c}\right)^2 + \left(\frac{2n-1}{b}\right)^2} \right]^{-1},$$

$$B_{n,m,k} = -64A \left[\pi^3 (2n-1)(2m-1)(2k-1) \left\{ \left(\frac{2m-1}{c}\right)^2 + \left(\frac{2n-1}{b}\right)^2 + \left(\frac{2k-1}{a}\right)^2 \right\} \right]^{-1}.$$

22. Решить уравнение Пуассона в цилиндре $0 \leq \rho \leq b$, $0 \leq \varphi \leq 2\pi$, $0 \leq z \leq h$

$$\Delta u = A,$$

$$\frac{\partial u}{\partial \rho} \Big|_{\rho=b} = 0, \quad u|_{z=0} = 0, \quad u|_{z=h} = \rho \cos \varphi.$$

Указание. $u = w + u_1$, $\Delta u_1 = 0$, $u_1|_{z=h} = \rho \cos(\varphi)$ и $u_1 = 0$ на остальной части границы; $u_1 = \cos(\varphi) \sum_{n=1}^{\infty} A_n J_1\left(\frac{\mu_n \rho}{b}\right) \operatorname{sh}\left(\frac{\mu_n z}{b}\right)$, где μ_n — последовательные положительные корни уравнения $J_1'(\mu) = 0$;

$$w = \sum_{n,m=1}^{\infty} B_{n,m} J_0\left(\frac{\mu_n \rho}{b}\right) \sin\left(\frac{m\pi z}{h}\right), \text{ где } B_{n,m} \text{ определяются из разложения}$$

$$A = - \sum_{n,m=1}^{\infty} B_{n,m} \left[\left(\frac{\mu_n}{b}\right)^2 + \left(\frac{m\pi}{h}\right)^2 \right] J_0\left(\frac{\mu_n \rho}{b}\right) \sin\left(\frac{m\pi z}{h}\right).$$

23. Решить смешанную задачу

$$\frac{\partial^2 u}{\partial x^2} + \frac{1}{x} \frac{\partial u}{\partial x} = \frac{\partial^2 u}{\partial t^2} + u, \quad 0 < x < 1,$$

$$|u|_{x=0} < \infty, \quad u|_{x=1} = \cos 2t + \sin 3t,$$

$$u|_{t=0} = \frac{J_0(x\sqrt{3})}{J_0(\sqrt{3})}, \quad \frac{\partial u}{\partial t} \Big|_{t=0} = \frac{3 J_0(2x\sqrt{2})}{J_0(2\sqrt{2})}.$$

$$\text{Ответ. } u = \frac{J_0(x\sqrt{3})}{J_0(\sqrt{3})} \cos 2t + \frac{J_0(2x\sqrt{2})}{J_0(2\sqrt{2})} \sin 3t.$$

24. Найти решение смешанной задачи

$$\frac{\partial^2 u}{\partial x^2} + \frac{1}{x} \frac{\partial u}{\partial x} = \frac{\partial^2 u}{\partial t^2} + \cos t, \quad 0 < x < 1,$$

$$|u|_{x=0} < \infty, \quad u|_{x=1} = 0,$$

$$u|_{t=0} = 1 - \frac{J_0(x)}{J_0(1)}, \quad \frac{\partial u}{\partial t}|_{t=0} = 0.$$

$$\text{Ответ. } u = \left[1 - \frac{J_0(x)}{J_0(1)} \right] \cos t.$$

25. Найти решение смешанной задачи

$$\frac{\partial^2 u}{\partial x^2} + \frac{1}{x} \frac{\partial u}{\partial x} = \frac{\partial^2 u}{\partial t^2} + \sin 3t, \quad 0 < x < 1,$$

$$|u|_{x=0} < \infty, \quad u|_{x=1} = 1,$$

$$u|_{t=0} = 1, \quad \frac{\partial u}{\partial t}|_{t=0} = \frac{1}{3} \left[1 - \frac{J_0(3x)}{J_0(3)} \right].$$

$$\text{Ответ. } u = 1 + \frac{1}{9} \sin 3t \left[1 - \frac{J_0(3x)}{J_0(3)} \right].$$

6. Задачи математической физики с непрерывным спектром. Интегральные преобразования

В тех случаях, когда применение метода Фурье приводит к совокупности частных решений, непрерывно зависящих от некоторого параметра, изменяющегося в заданном промежутке, рассматриваемый вопрос относится к классу задач математической физики с непрерывным спектром. Особенность данного случая, который встречается, как правило, при рассмотрении задач для бесконечных областей, заключается в том, что искомое решение строится из найденных частных решений путем интегрирования по параметру, то есть представляется в форме разложения в интеграл по собственным функциям.

На рис. 6.1 приведена таблица, классифицирующая соответствующие задачи Штурма–Лиувилля.




Рис. 6.1. Классификация задач Штурма-Лиувилля

В случае регулярной задачи Штурма–Лиувилля, мы имеем дискретный спектр, которому соответствует счетное множество собственных функций. В случае сингулярной задачи Штурма–Лиувилля спектр может быть непрерывным, дискретным или смешанным. Если спектр задачи непрерывный, то мы имеем дело с несчетным множеством собственных функций. В этом случае решение задачи получается в виде разложения в интеграл. Для определения коэффициентов такого разложения используются теоремы о разложении функции в интеграл.

Некоторые интегральные разложения, связанные с сингулярной задачей Штурма–Лиувилля

К числу наиболее простых и важных интегральных разложений, встречающихся в математической физике, относятся классическое разложение функции, заданной в промежутке $(-\infty, +\infty)$, в интеграл Фурье и его различные модификации (разложение функции, заданной в промежутке $(0, +\infty)$, в синус- и косинус-интеграл Фурье и другие разложения по тригонометрическим функциям), разложение функции, определенной в промежутке $(0, +\infty)$, в интеграл Ханкеля по цилиндрическим функциям и некоторые другие разложения.

I. Простейшие интегральные разложения связаны со следующей задачей Штурма–Лиувилля:

$$X'' + \lambda X = 0, \quad 0 < x < \infty,$$

$$\left. \begin{array}{l} \text{(I)} \quad X(0) = 0, \quad \text{либо} \\ \text{(II)} \quad X'(0) = 0, \quad \text{либо} \\ \text{(III)} \quad X'(0) - hX(0) = 0 \quad (h > 0) \end{array} \right\} \quad X|_{x \rightarrow \infty} - \text{ограничена.}$$

Для условий первого рода собственные значения удобно представить как

$$\lambda = \lambda_\nu = \nu^2, \quad 0 < \nu < \infty,$$

то есть, спектр задачи непрерывный $\lambda \in (0, +\infty)$. Собственные функции имеют вид:

$$X = X_\nu(x) = \sin(\nu x).$$

Легко проверить, что $\lambda = \lambda_0 = 0$ не является собственным значением.

Для условий второго рода ($\lambda \neq 0$) имеем:

$$\lambda = \lambda_\nu = \nu^2, \quad 0 < \nu < \infty.$$

Собственные функции:

$$X = X_\nu(x) = \cos(\nu x).$$

Как легко проверить, число $\lambda = \lambda_0 = 0$ является собственным значением, которому соответствует собственная функция $X_0(x) = 1$. Таким образом, в случае условий второго рода спектр задачи непрерывный $\lambda \in [0, +\infty)$.

Для условий третьего рода ($\lambda \neq 0$) будем иметь

$$\lambda = \lambda_\nu = \nu^2, \quad 0 < \nu < \infty,$$

$$X_\nu(x) = A \left[\cos(\nu x) + \frac{h}{\nu} \sin(\nu x) \right].$$

Здесь A — произвольная константа. Удобно положить

$$A = \frac{\nu}{\sqrt{\nu^2 + h^2}}.$$

Тогда условия первого и второго рода получаются из условий третьего рода как частные случаи. Очевидно, $\lambda = 0$ не является собственным числом задачи (за исключением случая $h = 0$).

Таким образом, окончательно

$$X = X_\nu(x) = \frac{\nu \cos(\nu x) + h \sin(\nu x)}{\sqrt{\nu^2 + h^2}}$$

— собственные функции задачи. Спектр — непрерывный $\lambda \in (0, +\infty)$.

Каждому из рассмотренных случаев соответствует своя теорема о разложении функции в интеграл.

Для условий первого рода это будет *синус-интеграл Фурье*:

$$f(x) = \int_0^{\infty} F(\nu) \sin(\nu x) d\nu, \quad 0 < x < +\infty, \quad (6.1)$$

при этом

$$F(\nu) = \frac{2}{\pi} \int_0^{\infty} f(x) \sin(\nu x) dx. \quad (6.2)$$

Для условий второго рода — *косинус-интеграл Фурье*:

$$f(x) = \int_0^{\infty} F(\nu) \cos(\nu x) d\nu, \quad 0 < x < +\infty,$$

при этом

$$F(\nu) = \frac{2}{\pi} \int_0^{\infty} f(x) \cos(\nu x) dx.$$

Для условий третьего рода — *обобщенный интеграл Фурье*:

$$f(x) = \int_0^{+\infty} F(\nu) \frac{\nu \cos(\nu x) + h \sin(\nu x)}{\sqrt{\nu^2 + h^2}} d\nu, \quad 0 < x < +\infty,$$

при этом

$$F(\nu) = \frac{2}{\pi} \int_0^{\infty} f(x) \frac{\nu \cos(\nu x) + h \sin(\nu x)}{\sqrt{\nu^2 + h^2}} dx.$$

Достаточные условия для применимости всех трех теорем — функция $f(x)$ должна удовлетворять условиям Дирихле в любом замкнутом промежутке на полуоси $(0, +\infty)$ и быть абсолютно интегрируемой на этой полуоси, то есть

- $f(x)$ — кусочно-непрерывна на $(0, +\infty)$ и имеет конечное число максимумов и минимумов на $[\alpha, \beta] \subset (0, +\infty)$;
- $\int_0^{\infty} |f(x)| dx$ сходится.

В точках разрыва интеграл Фурье сходится к значению

$$\frac{1}{2} [f(x-0) + f(x+0)].$$

Второе из сформулированных условий допускает обращение функции $f(x)$ в бесконечность, если только несобственный интеграл сходится.

II. Следующая сингулярная задача Штурма–Лиувилля:

$$X'' + \lambda X = 0, \quad -\infty < x < +\infty,$$

$$X|_{x \rightarrow \pm\infty} = O(1).$$

имеет непрерывный спектр $\lambda \in [0, +\infty)$:

$$\lambda = \lambda_\nu = \nu^2, \quad 0 \leq \nu < +\infty.$$

Собственные функции будут

$$X(x) = X_\nu(x) = A_\nu \cos(\nu x) + B_\nu \sin(\nu x).$$

В этом случае теорема разложения имеет вид:

$$f(x) = \int_0^{\infty} [M(\nu) \cos(\nu x) + N(\nu) \sin(\nu x)] d\nu, \quad -\infty < x < +\infty$$

— обычный интеграл Фурье, где

$$M(\nu) = \frac{1}{\pi} \int_{-\infty}^{+\infty} f(x) \cos(\nu x) dx,$$

$$N(\nu) = \frac{1}{\pi} \int_{-\infty}^{+\infty} f(x) \sin(\nu x) dx.$$

Достаточные условия для применимости теоремы разложения — функция $f(x)$ абсолютно интегрируема на всей оси и удовлетворяет условиям Дирихле в любом замкнутом промежутке, то есть

- $f(x)$ — кусочно-непрерывна на $(-\infty, +\infty)$ и имеет конечное число максимумов и минимумов $\forall x \in [\alpha, \beta] \subset (-\infty, +\infty)$;
- $\int_{-\infty}^{\infty} |f(x)| dx$ — сходится.

Можно записать *интеграл Фурье в комплексной форме*

$$f(x) = \int_{-\infty}^{+\infty} F(v) e^{-ivx} dv, \quad F(v) = \frac{1}{2\pi} \int_{-\infty}^{+\infty} f(x) e^{ivx} dx.$$

III. Рассмотрим еще одну часто встречающуюся в приложениях задачу:

$$(rR')' + \lambda rR = 0, \quad 0 < r < +\infty,$$

$$R|_{r \rightarrow 0} - \text{ограничена}; \quad R|_{r \rightarrow +\infty} - \text{ограничена}.$$

Здесь собственные значения будут

$$\lambda = \lambda_v = v^2, \quad 0 \leq v < \infty.$$

Собственные функции

$$R = R_v(r) = J_0(vr).$$

Спектр задачи непрерывный $\lambda \in [0, +\infty)$.

Теоремой разложения в этом случае служит формула *интеграла Фурье–Бесселя*:

$$f(r) = \int_0^{\infty} F(v) J_0(vr) dv, \quad 0 < r < +\infty,$$

$$F(v) = v \int_0^{\infty} f(r) r J_0(vr) dr.$$

Достаточные условия для применимости теоремы разложения: функция $\sqrt{r}f(r)$ абсолютно интегрируема в интервале $(0, +\infty)$ и удовлетворяет условиям Дирихле в любом замкнутом промежутке из этого интервала, то есть

- $f(r)$ кусочно-непрерывна на $(0, +\infty)$ и имеет конечное число максимумов и минимумов $\forall r \in [\alpha, \beta] \subset (0, +\infty)$;
- $\int_0^{\infty} \sqrt{r} |f(r)| dr$ сходится.

Метод интегральных преобразований

Для решения задач математической физики с непрерывным спектром во многих случаях может быть с успехом использован метод интегральных преобразований. Под интегральным преобразованием функции $f(x)$ понимается значение определенного интеграла, взятого по заданному промежутку¹ от произведения $f(x)$ на ядро преобразования, представляющее собой функцию переменной x и некоторого параметра, который может принимать произвольные значения в заданной вещественной или комплексной области.

Интегральные преобразования, встречающиеся в математической физике, могут быть условно разделены на «вещественные» и «комплексные», в зависимости от значений, принимаемых параметром преобразования.

Пусть задана некоторая функция $K(x, v)$, где $a < x < \infty$, $b < v < \infty$. Будем предполагать эту функцию непрерывной от x и v в указанной области. Пусть $f(x) \in A$ — функция вещественной переменной, (A — некоторый класс функций). Если для каждой функции класса A интеграл

$$F(v) = \int_a^{\infty} K(x, v) f(x) dx \quad (6.3)$$

сходится, то в этом случае говорят, что определено *интегральное преобразование* $F(v)$ от функции $f(x)$ на классе A . При этом функция $K(x, v)$ называется *ядром* интегрального преобразования.

Во многих случаях существует обратная зависимость для (6.3). Она имеет вид

$$f(x) = \int_b^{\infty} M(x, v) F(v) dv. \quad (6.4)$$

Функция $M(x, v)$ называется *ядром обратного преобразования*. Формула (6.4) — обращение преобразования (6.3). Конкретная структура ядра $M(x, v)$ зависит от ядра $K(x, v)$ и пределов изменения переменных. Как правило, формула (6.4) имеет место в некотором классе B (не на всем классе A : $B \subset A$), то есть $f(x) \in B$.

Обе формулы (6.3) и (6.4) можно объединить в одну

$$f(x) = \int_a^{\infty} M(x, v) dv \int_a^{\infty} K(y, v) f(y) dy, \quad f(x) \in B, \quad a < x < \infty. \quad (6.5)$$

¹ В задачах математической физики этот промежуток совпадает с интервалом изменения той переменной, по функциям которой ведется разложение.

Формула (6.5) — разложение функции $f(x)$ по функциям $M(x, v)$. Часто случается, что формулы (6.3) и (6.4) взаимны.

К группе вещественных преобразований, общий вид которых может быть представлен в форме (6.3), относятся, например, следующие преобразования.

Синус-преобразование Фурье:

$$F(v) = \int_0^{\infty} f(x) \sin(vx) dx, \quad 0 < v < \infty \quad (6.6)$$

Формула обращения синус-преобразования Фурье:

$$f(x) = \frac{2}{\pi} \int_0^{\infty} F(v) \sin(vx) dv, \quad 0 < x < \infty. \quad (6.7)$$

Здесь $K(x, v) = \sin(vx)$, $M(x, v) = 2 \sin(vx)/\pi$. Обратим внимание, что теперь, в отличие от формул (6.1) и (6.2), числовой коэффициент $2/\pi$ стоит перед интегралом для функции $F(v)$. Положение этого коэффициента, вообще говоря, неважно. Иногда, чтобы сделать формулы типа (6.6) и (6.7) симметричными, ставится числовой коэффициент $\sqrt{2/\pi}$ перед обоими интегралами в этих формулах. Аналогичное замечание справедливо и для других преобразований.

В качестве класса A можно взять класс кусочно-непрерывных и абсолютно интегрируемых функций на $(0, +\infty)$. Класс B — класс таких же функций, но имеющих конечное число максимумов и минимумов на любом $[\alpha, \beta] \subset (0, +\infty)$.

Косинус-преобразование Фурье:

$$F(v) = \int_0^{\infty} f(x) \cos(vx) dx, \quad 0 < v < \infty.$$

Формула обращения

$$f(x) = \frac{2}{\pi} \int_a^{\infty} F(v) \cos(vx) dv, \quad 0 < x < \infty.$$

Обобщенное преобразование Фурье:

$$F(v) = \int_a^{\infty} f(x) \frac{v \cos(vx) + h \sin(vx)}{\sqrt{v^2 + h^2}} dx, \quad 0 < v < \infty.$$

Формула обращения

$$f(x) = \frac{2}{\pi} \int_a^{\infty} F(v) \frac{v \cos(vx) + h \sin(vx)}{\sqrt{v^2 + h^2}} dv, \quad 0 < x < \infty.$$

Классическое преобразование Фурье:

$$F(v) = \int_{-\infty}^{\infty} f(x) e^{ivx} dx, \quad -\infty < v < \infty.$$

Формула обращения:

$$f(x) = \frac{1}{2\pi} \int_{-\infty}^{\infty} F(v) e^{-ivx} dv, \quad -\infty < x < \infty.$$

Ядра классического преобразования Фурье имеют структуру

$$K(x, v) = e^{ivx}, \quad M(x, v) = \frac{1}{2\pi} e^{-ivx}.$$

Преобразование Фурье–Бесселя:

$$F(v) = \int_a^{\infty} f(r) r J_0(vr) dr, \quad 0 < v < \infty.$$

Формула обращения

$$f(r) = \int_a^{\infty} F(v) v J_0(vr) dv, \quad 0 < r < \infty.$$

Здесь ядра имеют вид $K(r, v) = r J_0(vr)$, $M(x, v) = v J_0(vr)$. Класс функций, для которых справедливо преобразование, определяется теоремой разложения в интеграл Фурье–Бесселя.

Типичными представителями другой группы преобразований, которые могут быть представлены в виде

$$F(p) = \int_a^{\infty} K(x, p) f(x) dx$$

(a — заданное число, $p = \sigma + i\tau$ — параметр, изменяющийся в некоторой области D плоскости комплексного переменного, K — ядро преобразования), являются:

- преобразование Лапласа: $K(x, p) = e^{-xp}$, $a = 0$, D — полуплоскость, лежащая правее некоторой прямой $\sigma = \sigma_1$, параллельной мнимой оси;
- преобразование Меллина: $K(x, p) = x^{p-1}$, $a = 0$, D — полоса, заключенная между параллельными прямыми $\sigma = \sigma_1$ и $\sigma = \sigma_2$.

С помощью интегральных преобразований можно решать неоднородные задачи математической физики по схеме метода Гринберга. Применение интегрального преобразования к дифференциальному уравнению в частных производных временно исключает одну из независимых переменных, вследствие чего интегрирование заданного дифференциального уравнения в частных производных сводится к интегрирова-

нию уравнения в частных производных, содержащих на единицу меньше независимых переменных, чем заданное уравнение. Если исходное уравнение с двумя независимыми переменными, то применение интегрального преобразования сводит задачу к интегрированию обыкновенного дифференциального уравнения.

Решение преобразованного таким образом уравнения является функцией от v и остальных переменных. После того как указанное решение получено, требуется «обратить» его, чтобы восстановить «утраченную» переменную. Этот процесс обращения сводится, по существу, к решению интегрального уравнения относительно функции $f(x)$; функция $F(v)$ предполагается известной. Формальное решение этого интегрального уравнения дается формулой обращения.

Достоинствами метода интегральных преобразований являются возможность применения метода к однородным и неоднородным задачам, упрощение выкладок и разделение принципиальной и чисто вычислительной части решения, а также возможность построения операционного исчисления для данного ядра путем составления таблиц прямых и обратных преобразований различных часто встречающихся в приложениях функций.

Преобразование Фурье

Широко используемым в теоретических исследованиях и в приложениях является интегральное преобразование Фурье, поэтому напомним здесь некоторые основные свойства этого преобразования.

Пусть $f(x)$ — произвольная функция, определенная на интервале $(-\infty, +\infty)$ и удовлетворяющая условиям:

1. $f(x)$ — кусочно-непрерывна на $(-\infty, +\infty)$;
2. $f(x)$ абсолютно интегрируема на всей числовой оси, то есть $\int_{-\infty}^{+\infty} |f(x)| dx$ имеет конечное значение.

Будем говорить, что тогда $f(x) \in A$. При таких условиях функция $f(x)$ может быть представлена в виде разложения в интеграл Фурье

$$f(x) = \frac{1}{2\pi} \int_{-\infty}^{\infty} e^{-isx} \int_{-\infty}^{\infty} f(\xi) e^{is\xi} d\xi, \quad -\infty < x < +\infty,$$

причем в точке разрыва первого рода $x = c$ левая часть формулы должна быть заменена полусуммой $[f(c-0) + f(c+0)]/2$.

Заметим еще, что последняя формула может быть записана в вещественной форме

$$f(x) = \frac{1}{\pi} \int_0^{\infty} \left[\cos(sx) \int_{-\infty}^{\infty} f(\xi) \cos(s\xi) d\xi + \sin(sx) \int_{-\infty}^{\infty} f(\xi) \sin(s\xi) d\xi \right] ds.$$

Преобразованием Фурье от функции $f(x)$ называется интеграл

$$\tilde{f}(s) = \int_{-\infty}^{\infty} f(x) e^{isx} dx, \quad -\infty < s < +\infty.$$

Для функций, удовлетворяющих перечисленным условиям, преобразование Фурье всегда существует.

Формула обращения имеет вид (в точках непрерывности)

$$f(x) = \frac{1}{2\pi} \int_{-\infty}^{\infty} \tilde{f}(s) e^{-isx} ds, \quad -\infty < x < +\infty.$$

Эта формула справедлива для функций $f(x) \in B$, удовлетворяющих условиям:

1. $f(x)$ — кусочно-непрерывна и имеет конечное число максимумов и минимумов в любом замкнутом промежутке $[a, b] \subset (-\infty, +\infty)$;
2. $f(x)$ абсолютно интегрируема на всей числовой оси, то есть $\int_{-\infty}^{+\infty} |f(x)| dx$ имеет конечное значение.

Преобразование Лапласа

Преобразование Лапласа также имеет широкую область приложений, поэтому напомним здесь некоторые основные факты, связанные с этим преобразованием.

Пусть задана некоторая функция $f(t)$, $t \in (0, +\infty)$. Рассмотрим интеграл (*интеграл Лапласа*)

$$\bar{f}(p) = \int_0^{+\infty} f(t) e^{-pt} dt, \quad (6.8)$$

где p — некоторое комплексное число, $\operatorname{Re}(p) > a$ (a — любое).

Если интеграл (6.8) сходится при некотором конкретном значении a для данной функции $f(t)$, то говорят, что этот интеграл определяет преобразование Лапласа от функции $f(t)$. В этом случае комплексная, вообще говоря, функция $f(t)$ действительного аргумента t называется *оригиналом* или *начальной функцией*, а $\bar{f}(p)$ — *изображением* или *трансформантой Лапласа*.

Будем говорить, что $f(t) \in M$, если

- функция $f(t)$ кусочно-непрерывна на $(0, +\infty)$;

- интеграл $\int_0^{+\infty} |f(t)| e^{-ct} dt$ сходится при соответствующем значении c .

Можно показать, что для функций класса M преобразование Лапласа заведомо существует.

ПРИМЕЧАНИЕ

Мы можем ограничить класс рассматриваемых функций кусочно-непрерывными функциями $f(t)$ вещественной переменной t , определенными при $t > 0$ и принимаемыми равными нулю при $t < 0$.

Из класса кусочно-непрерывных функций мы выделяем подкласс функций, характеризуемых *определенным порядком роста* при весьма больших значениях аргумента t . Мы предполагаем, что можно указать такие, независимые от t числа L и c , что каково бы ни было t $|f(t)| < Le^{ct}$, то есть при любом $t > 0$ модуль $f(t)$ растет медленнее, чем некоторая экспоненциальная функция.

Таким образом, функция $f(t)$ называется оригиналом, если она удовлетворяет следующим условиям:

1. $f(t)$ — дифференцируемая достаточное число раз на всей оси t , кроме точек, где она и ее производные претерпевают разрыв первого рода, причем число таких точек в конечном промежутке конечно;
2. для всех отрицательных t функция $f(t) = 0$;
3. $f(t)$ возрастает не быстрее показательной функции.

ПРИМЕЧАНИЕ

В таблицах преобразований Лапласа неявно подразумевается, что функции $f(t)$ равны нулю при $t < 0$. Значения этих функций приводятся только при $t > 0$. Это обстоятельство может привести к серьезным ошибкам, если не обращать на него внимания.

Формула обращения

Будем считать, что $f(t) \in N \subset M$. Мы скажем, что $f(t) \in N$, если

1. функция $f(t)$ кусочно-непрерывна в интервале $(0, +\infty)$;

2. интеграл $\int_0^{+\infty} |f(t)| e^{-ct} dt$ сходится при соответствующем выборе числа c ;
3. функция $f(t) e^{-at}$, $a > c$ имеет конечное число максимумов и минимумов на любом замкнутом промежутке $[\alpha, A] \subset (0, +\infty)$.

Формула обращения имеет вид

$$f(t) = \frac{1}{2\pi i} \int_{a-i\infty}^{a+i\infty} \bar{f}(p) e^{pt} dp . \quad (6.9)$$

Предполагается, что $a > c$, $t > 0$ — точки непрерывности. Здесь интеграл в (6.9) понимается в смысле главного значения, то есть

$$\int_{a-i\infty}^{a+i\infty} \bar{f}(p) e^{pt} dp = \lim_{T \rightarrow \infty} \int_{a-iT}^{a+iT} \bar{f}(p) e^{pt} dp .$$

Формула (6.9) называется *формулой Римана–Меллина*, а интеграл справа — *интегралом Римана–Меллина*.

Неоднородные задачи с непрерывным спектром

Рассмотрим технику применения метода интегральных преобразований на примере следующей задачи: найти функцию $u = u(x, y)$, удовлетворяющую следующим условиям: уравнению

$$L_x(u) + M_y(u) = F(x, y), \quad a < x < \infty, \quad c < y < d ,$$

где L_x , M_y — линейные дифференциальные операторы

$$L_x(u) = \frac{1}{r(x)} \left[\frac{\partial}{\partial x} \left(p(x) \frac{\partial u}{\partial x} \right) - q(x)u \right],$$

$$M_y(u) = A(y) \frac{\partial^2 u}{\partial y^2} + B(y) \frac{\partial u}{\partial y} + C(y)u .$$

Если $x = a$ — регулярная граница, то удовлетворяются условия одного из следующих типов:

$$u|_{x=a} = f(y) ,$$

$$\frac{\partial u}{\partial x} \Big|_{x=a} = f(y) ,$$

$$\frac{\partial u}{\partial x} - hu \Big|_{x=a} = f(y), \quad h > 0 .$$

На сингулярной границе выполняется условие: $u|_{x \rightarrow \infty} = O(1)$.

Если $x = a$ — сингулярная граница, то $u|_{x \rightarrow a} = O(1)$, $u|_{x \rightarrow \infty} = O(1)$. Далее, выполняются некоторые условия по переменной y , которые зависят от типа уравнения.

Соответствующая задача Штурма–Лиувилля имеет вид:

$$(pX')' + (\lambda r - q)X = 0.$$

Если $x = a$ — регулярная граница, то

$$X|_{x=a} = 0,$$

либо

$$\left. \frac{dX}{dx} \right|_{x=a} = 0,$$

либо

$$\left. \frac{dX}{dx} - hX \right|_{x=a} = 0;$$

и

$$X|_{x \rightarrow \infty} = O(1).$$

Если $x = a$ — сингулярная граница, то

$$X|_{x \rightarrow a} = O(1), \quad X|_{x \rightarrow \infty} = O(1).$$

Будем считать, что сформулированная задача Штурма–Лиувилля имеет непрерывный спектр собственных значений, которые обозначим $\lambda = \lambda_\nu = \nu^2$; а собственные функции — $X = X_\nu(x)$, $0 < \nu < \infty$.

Поступим по аналогии с методом Гринберга:

$$\begin{aligned} \int_a^\infty X_\nu \left[\frac{\partial}{\partial x} \left(p(x) \frac{\partial u}{\partial x} \right) - q(x)u \right] dx + \int_a^\infty M_y(u) r X_\nu dx &= \int_a^\infty F(x, y) r X_\nu dx \Rightarrow \\ p \left(X_\nu \frac{\partial u}{\partial x} - X'_\nu u \right) \Big|_{x=a}^{x=\infty} + \int_a^\infty u \underbrace{\left[(pX'_\nu)' - qX_\nu \right]}_{=-\lambda_\nu r X_\nu} dx + M_y \left(\int_a^\infty u r X_\nu dx \right) &= \\ = \underbrace{\int_a^\infty F(x, y) r X_\nu dx}_{=\bar{F}_\nu(y)}. \end{aligned}$$

Введем в рассмотрение интегральное преобразование

$$\bar{u}_v = \int_a^\infty urX_v dx .$$

Тогда наше уравнение будет иметь вид

$$M_y(\bar{u}_v) - \lambda_v \bar{u}_v = \bar{F}_v(y) - G_v(y) ,$$

где

$$G_v(y) = p \left(X_v \frac{\partial u}{\partial x} - X'_v u \right) \Big|_{x=a}^{x=\infty} .$$

Если $x = a$ — регулярная граница, то правая часть в полученном уравнении при $x = a$ — известная функция. Наложим некоторые дополнительные условия при $x = \infty$. Будем считать, что

$$p \left(X_v \frac{\partial u}{\partial x} - X'_v u \right) \Big|_{x=\infty} = 0 .$$

При таких условиях $G_v(y)$ — известная функция.

Если $x = a$ — сингулярная граница, то потребуем, чтобы

$$p \left(X_v \frac{\partial u}{\partial x} - X'_v u \right) \Big|_{x=a} = 0, \quad p \left(X_v \frac{\partial u}{\partial x} - X'_v u \right) \Big|_{x=\infty} = 0 .$$

При таких дополнительных условиях метод интегральных преобразований применим. Для полной постановки задачи для полученного обыкновенного дифференциального уравнения относительно \bar{u}_v необходимо еще трансформировать условия по переменной y . Решив это уравнение с соответствующими условиями по переменной y , найдем функцию \bar{u}_v . Затем функцию u найдем по формуле обращения

$$u(x, y) = \int_a^\infty M(x, v) \bar{u}_v(y) dv ,$$

где $M(x, v)$ — ядро обратного преобразования; ядро прямого преобразования имеет вид $K(x, v) = r(x) X_v(x)$.

Может оказаться так, что сформулированные дополнительные условия неприемлемы для рассматриваемой задачи. Тогда метод интегральных преобразований тоже не применим.

Примеры решения типовых задач

Задача 6.1. Изучить напряженное состояние упругой полуплоскости ($y \geq 0$), вызванное произвольными усилиями, приложенными на границе:

$$\sigma_y|_{y=0} = f(x), \quad \tau_{xy}|_{y=0} = g(x). \quad (*)$$

Рассмотреть частный случай, когда в начале координат к полуплоскости приложена сосредоточенная сила \vec{P} с составляющими $P_x = 0$, $P_y = P$ — задача Фламана.

Решение. Решение плоской задачи теории упругости может быть сведено к интегрированию бигармонического уравнения для функции напряжений Эри

$$\Delta^2 \Phi = 0 \Leftrightarrow \frac{\partial^4 \Phi}{\partial x^4} + 2 \frac{\partial^4 \Phi}{\partial x^2 \partial y^2} + \frac{\partial^4 \Phi}{\partial y^4} = 0,$$

причем напряжения выражаются через функцию Эри по формулам (в предположении отсутствия объемных нагрузок)

$$\sigma_x = \frac{\partial^2 \Phi}{\partial y^2}, \quad \sigma_y = \frac{\partial^2 \Phi}{\partial x^2}, \quad \tau_{xy} = -\frac{\partial^2 \Phi}{\partial xy}.$$

Решаем задачу в Maple. Определяем уравнение и дополнительные условия

```
> pde :=
VectorCalculus[Laplacian] (VectorCalculus[Laplacian]
(Phi(x, y), 'cartesian'[x, y]), 'cartesian'[x, y])=0;
pde := ( (∂⁴/∂x⁴ Φ(x, y)) + 2 (∂⁴/∂y² ∂x² Φ(x, y)) + (∂⁴/∂y⁴ Φ(x, y)) ) = 0
```

Область изменения переменных: $-\infty < x < \infty$, $y \geq 0$. Поэтому применяем классическое преобразование Фурье по переменной x ; в системе Maple преобразование Фурье определяется по формуле

$$\varphi(y) = \int_{-\infty}^{+\infty} \Phi(x, y) e^{-isx} dx.$$

```
> inttrans[fourier] (pde, x, s);
ode := subs (fourier (Phi(x, y), x, s) = phi(y), %);
( (∂⁴/∂y⁴ fourier (Φ(x, y), x, s)) + s⁴ fourier (Φ(x, y), x, s)
- 2 s² (∂²/∂y² fourier (Φ(x, y), x, s)) ) = 0
ode := ( (d⁴/dy⁴ φ(y)) + s⁴ φ(y) - 2 s² (d²/dy² φ(y)) ) = 0
```

Таким образом, мы получили обыкновенное дифференциальное уравнение для трансформанты Фурье

$$\frac{d^4\varphi}{dy^4} - 2s^2 \frac{d^2\varphi}{dy^2} + s^4\varphi = 0.$$

Определим теперь формулы для вычисления напряжений — напряжение σ_x :

```
> sigma[1]:=diff(Phi(x,y),y$2);
lprint(`трансформанта Фурье этого напряжения:`);
S[1]:=inttrans[fourier](%,x,s);
S[1]:=subs(inttrans[fourier](Phi(x,y),x,s)=phi(y),S[1]);
```

$$\sigma_1 := \frac{\partial^2}{\partial y^2} \Phi(x, y)$$

```
`трансформанта Фурье этого напряжения:`
```

$$S_1 := \frac{\partial^2}{\partial y^2} \text{fourier}(\Phi(x, y), x, s)$$

$$S_1 := \frac{d^2}{dy^2} \phi(y)$$

напряжение σ_y :

```
> sigma[2]:=diff(Phi(x,y),x$2);
lprint(`трансформанта Фурье этого напряжения:`);
S[2]:=inttrans[fourier](%,x,s);
S[2]:=subs(inttrans[fourier](Phi(x,y),x,s)=phi(y),S[2]);
```

$$\sigma_2 := \frac{\partial^2}{\partial x^2} \Phi(x, y)$$

```
`трансформанта Фурье этого напряжения:`
```

$$S_2 := -s^2 \text{fourier}(\Phi(x, y), x, s)$$

$$S_2 := -s^2 \phi(y)$$

напряжение τ_{xy} :

```
> tau:=-diff(Phi(x,y),x,y);
lprint(`трансформанта Фурье этого напряжения:`);
Tau:=inttrans[fourier](%,x,s);
Tau:=subs(inttrans[fourier](Phi(x,y),x,s)=phi(y),Tau);
```

$$\tau := -\left(\frac{\partial^2}{\partial y \partial x} \Phi(x, y)\right)$$

```
`трансформанта Фурье этого напряжения:`
```

$$T := -I s \left(\frac{\partial}{\partial y} \text{fourier}(\Phi(x, y), x, s) \right)$$

$$T := -I s \left(\frac{d}{dy} \phi(y) \right)$$

Находим общее выражение для трансформанты Фурье функции напряжений Эри

> dsolve(ode, phi(y)); assign(%):

$$\phi(y) = _C1 e^{(s y)} + _C2 e^{(s y)} y + _C3 e^{(-s y)} + _C4 e^{(-s y)} y$$

Учитывая ограниченность на бесконечности, получаем

> phi:=unapply(subs({_C1=0, _C2=0, s=abs(s)}, phi(y)), y);

$$\phi := y \rightarrow _C3 e^{(-|s|y)} + _C4 e^{(-|s|y)} y$$

То есть, мы получили общее выражение для трансформанты Фурье, ограниченное на бесконечности, в виде

$$\varphi(y) = C_3 e^{-|s|y} + C_4 y e^{-|s|y},$$

где C_3 и C_4 — произвольные постоянные. Определяем эти произвольные постоянные из граничных условий (*), предварительно трансформировав их с помощью преобразования Фурье

> eq1:=simplify(subs(y=0, S[2]))=F;

> eq2:=expand(simplify(subs(y=0, Tau)))=G;

$$eq1 := -_C3 s^2 = F$$

$$eq2 := s _C3 |s| I - s _C4 I = G$$

> _C3:=solve(eq1, _C3); _C4:=solve(eq2, _C4);

$$_C3 := -\frac{F}{s^2}$$

$$_C4 := \frac{(F|s|I + G s) I}{s^2}$$

Здесь мы обозначили через F и G трансформанты Фурье функций f и g соответственно.

> collect(subs({F=F(s), G=G(s)}, phi(y)), G(s));

phi:=unapply(collect(%, F(s)), y);

$$\frac{e^{(-|s|y)} y G(s) I}{s} - \frac{F(s) e^{(-|s|y)}}{s^2} - \frac{F(s) |s| e^{(-|s|y)} y}{s^2}$$

$$\phi := y \rightarrow \left(-\frac{e^{(-|s|y)}}{s^2} - \frac{|s| e^{(-|s|y)} y}{s^2} \right) F(s) + \frac{e^{(-|s|y)} y G(s) I}{s}$$

Итак, трансформанту Фурье функции Эри мы получили в виде

$$\varphi(y) = \frac{y e^{-|s|y} G(s) i}{s} - \left(\frac{e^{-|s|y}}{s^2} + \frac{e^{-|s|y} y |s|}{s^2} \right) F(s), \quad i = \sqrt{-1}.$$

Определяем теперь напряжения. Напряжение σ_x :

```
> inttrans[invfourier](S[1], s, x);
```

$$\text{invfourier}\left(\frac{|s|^2 e^{(-s|y)} F(s)}{s^2}, s, x\right) + y \text{invfourier}\left(\frac{|s|^2 e^{(-s|y)} G(s)}{s}, s, x\right) I$$

$$- 2 I \text{invfourier}\left(\frac{|s| e^{(-s|y)} G(s)}{s}, s, x\right) - y \text{invfourier}\left(\frac{|s|^3 e^{(-s|y)} F(s)}{s^2}, s, x\right)$$

```
> sigma[x]:=convert(%,int);
```

$$\sigma_x := \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \frac{|s|^2 e^{(-s|y)} F(s) e^{(s x I)}}{s^2} ds \right) + \frac{1}{2} \frac{I y}{\pi} \int_{-\infty}^{\infty} \frac{|s|^2 e^{(-s|y)} G(s) e^{(s x I)}}{s} ds$$

$$+ \frac{-I}{\pi} \int_{-\infty}^{\infty} \frac{|s| e^{(-s|y)} G(s) e^{(s x I)}}{s} ds - \frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} \frac{|s|^3 e^{(-s|y)} F(s) e^{(s x I)}}{s^2} ds \right)$$

%1 := s = -∞ .. ∞

Сразу же вычислим это напряжение в частном случае задачи Фламана

```
> assume(y>0): 'sigma[x]'=subs({G(s)=0, F(s)=-
P}, sigma[x]); 'sigma[x]'=simplify(rhs(%));
```

$$\sigma_x = \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} - \frac{|s|^2 e^{(-s|y\sim)} P e^{(s x I)}}{s^2} ds \right) - \frac{1}{2} \left(\frac{y\sim}{\pi} \int_{-\infty}^{\infty} - \frac{|s|^3 e^{(-s|y\sim)} P e^{(s x I)}}{s^2} ds \right)$$

$$+ \frac{1}{2} \frac{I y\sim}{\pi} \int_{-\infty}^{\infty} 0 ds + \frac{-I}{\pi} \int_{-\infty}^{\infty} 0 ds$$

%1 := s = -∞ .. ∞

$$\sigma_x = \frac{2 P y\sim x^2}{\pi (-x^2 + 2 I y\sim x + y\sim^2) (-y\sim^2 + 2 I y\sim x + x^2)}$$

Преобразуем результат, отделив вещественную и мнимую части

```
> 'sigma[x]'=combine(simplify(evalc(Re(rhs(%)))));
simplify(evalc(Im(rhs(%))));
```


$$\sigma_x = -\frac{2 y \sim P x^2}{\pi (y \sim^4 + 2 x^2 y \sim^2 + x^4)}$$

0

Итак, в частном случае задачи Фламана имеем

$$\sigma_x = -\frac{2 P y x^2}{\pi (x^2 + y^2)^2}.$$

Напряжение σ_y :

```
> inttrans[invfourier](S[2], s, x);
```

$$-y \left(\frac{\partial}{\partial x} \text{invfourier}(\mathbf{e}^{(-s|y)} G(s), s, x) \right) + \text{invfourier}(\mathbf{e}^{(-s|y)} F(s), s, x)$$

$$+ y \text{invfourier}(\mathbf{e}^{(-s|y)} F(s) |s|, s, x)$$

```
> sigma['y']:=convert(%,int);
```

$$\sigma_y := -\frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} \mathbf{e}^{(-s|y)} G(s) s \mathbf{e}^{(s x I)} I ds \right) + \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \mathbf{e}^{(-s|y)} F(s) \mathbf{e}^{(s x I)} ds \right)$$

$$+ \frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} \mathbf{e}^{(-s|y)} F(s) |s| \mathbf{e}^{(s x I)} ds \right)$$

Задача Фламана:

```
> 'sigma[y]'=subs({G(s)=0, F(s)=-P}, sigma['y']);
```

```
'sigma[y]'=simplify(rhs(%));
```

$$\sigma_y = \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} -\mathbf{e}^{(-s|y \sim)} P \mathbf{e}^{(s x I)} ds \right) - \frac{1}{2} \left(\frac{y \sim}{\pi} \int_{-\infty}^{\infty} 0 ds \right)$$

$$+ \frac{1}{2} \left(\frac{y \sim}{\pi} \int_{-\infty}^{\infty} -\mathbf{e}^{(-s|y \sim)} P |s| \mathbf{e}^{(s x I)} ds \right)$$

$$\sigma_y = \frac{2 P y \sim^3}{\pi (-x^2 + 2 I y \sim x + y \sim^2) (-y \sim^2 + 2 I y \sim x + x^2)}$$

```
> 'sigma[y]'=combine(simplify(evalc(Re(rhs(%)))));
```

```
simplify(evalc(Im(rhs(%))));
```

$$\sigma_y = -\frac{2 y \sim^3 P}{\pi (y \sim^4 + 2 x^2 y \sim^2 + x^4)}$$

0

Итак, в частном случае задачи Фламана имеем

$$\sigma_y = -\frac{2Py^3}{\pi(x^2 + y^2)^2}.$$

Напряжение τ_{xy} :

```
> inttrans[invfourier](Tau, s, x);
```

$$\left(y \operatorname{invfourier}(|s| e^{(-s|y)} G(s), s, x) I - \operatorname{invfourier}(e^{(-s|y)} G(s), s, x) I - y \operatorname{invfourier}\left(\frac{e^{(-s|y)} |s|^2 F(s)}{s}, s, x\right) \right) I$$

```
> tau:=convert(%,int);
```

$$\tau := \left(\frac{1}{2} I y \int_{-\infty}^{\infty} |s| e^{(-s|y)} G(s) e^{(sxI)} ds + \frac{-1}{2} I \int_{-\infty}^{\infty} e^{(-s|y)} G(s) e^{(sxI)} ds - \frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} \frac{e^{(-s|y)} |s|^2 F(s) e^{(sxI)}}{s} ds \right) \right) I$$

Задача Фламана:

```
> 'tau'=subs({G(s)=0, F(s)=-P}, tau); 'tau'=simplify(rhs(%));
```

$$\tau = \left(\frac{-1}{2} I \int_{-\infty}^{\infty} 0 ds + \frac{1}{2} I y \int_{-\infty}^{\infty} 0 ds - \frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} -\frac{e^{(-s|y)} |s|^2 P e^{(sxI)}}{s} ds \right) \right) I$$

$$\tau = \frac{2 y^2 P x}{\pi (-x^2 + 2 I y x + y^2) (-y^2 + 2 I y x + x^2)}$$

```
> 'tau'=combine(simplify(evalc(Re(rhs(%)))));
```

```
simplify(evalc(Im(rhs(%))));
```

$$\tau = -\frac{2 y^2 P x}{\pi (y^4 + 2 x^2 y^2 + x^4)}$$

0

Итак, в частном случае задачи Фламана имеем

$$\tau_{xy} = -\frac{2Py^2x}{\pi(x^2 + y^2)^2}.$$

Полученные общие формулы для напряжений можно упростить. Выполним все необходимые преобразования в Maple.

Преобразование формулы для напряжения σ_x .

> sigma [x];

$$\frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \frac{|s|^2 e^{(-s|y)} F(s) e^{(sxI)}}{s^2} ds \right) + \frac{1}{2} I y \int_{-\infty}^{\infty} \frac{|s|^2 e^{(-s|y)} G(s) e^{(sxI)}}{s} ds$$

$$+ \frac{-I}{\pi} \int_{-\infty}^{\infty} \frac{|s| e^{(-s|y)} G(s) e^{(sxI)}}{s} ds - \frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} \frac{|s|^3 e^{(-s|y)} F(s) e^{(sxI)}}{s^2} ds \right)$$

%1 := s = -∞ .. ∞

Подставим в эту формулу выражения F (s) и G (s) через интегралы

> F (s) := convert (fourier (f (xi) , xi , s) , int) ;

G (s) := convert (fourier (g (xi) , xi , s) , int) ;

$$F(s) := \int_{-\infty}^{\infty} f(\xi) e^{(-I\xi s)} d\xi$$

$$G(s) := \int_{-\infty}^{\infty} g(\xi) e^{(-I\xi s)} d\xi$$

Теперь напряжение σ_x имеет вид

> sigma [x];

$$\frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \frac{|s|^2 e^{(-s|y)} e^{(sxI)}}{s^2} \int_{-\infty}^{\infty} f(\xi) e^{(-I\xi s)} d\xi ds \right)$$

$$+ \frac{1}{2} I y \int_{-\infty}^{\infty} \frac{|s|^2 e^{(-s|y)} e^{(sxI)}}{s} \int_{-\infty}^{\infty} g(\xi) e^{(-I\xi s)} d\xi ds$$

$$\begin{aligned}
& + \frac{-I}{\pi} \int_{-\infty}^{\infty} \frac{|s| e^{(-s|y)} e^{(sxI)}}{s} \int_{-\infty}^{\infty} g(\xi) e^{(-I\xi s)} d\xi ds \\
& - \frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} \frac{|s|^3 e^{(-s|y)} e^{(sxI)}}{s^2} \int_{-\infty}^{\infty} f(\xi) e^{(-I\xi s)} d\xi ds \right)
\end{aligned}$$

%1 := -∞ .. ∞

Преобразуем отдельно интегралы, содержащие функцию $f(\xi)$ и интегралы, содержащие функцию $g(\xi)$. Объединим в один интеграл и поменяем порядок интегрирования в интегралах с функцией $f(\xi)$ (предполагаем законной такую операцию)

> sx1:=1/2*1/Pi*Int(f(xi)*Int(exp(-abs(s)*y+s*(x-xi)*I)*(1-abs(s)*y),s=-infinity..infinity),xi=-infinity..infinity);

$$sx1 := \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} f(\xi) \int_{-\infty}^{\infty} e^{(-s|y+s(x-\xi)I)} (1-|s|y) ds d\xi \right)$$

Выделяем внутренний интеграл и вычисляем его

> sx1_in:=Int(exp(-abs(s)*y+s*(x-xi)*I)*(1-abs(s)*y),s=-infinity..infinity);

$$sx1_in := \int_{-\infty}^{\infty} e^{(-s|y+s(x-\xi)I)} (1-|s|y) ds$$

> assume(y>0):

> sx1_in:=simplify(value(sx1_in));

$$sx1_in := -4(x-\xi)^2 y \sim / ((y \sim^2 + 2Ix y \sim - 2I\xi y \sim - x^2 + 2x\xi - \xi^2) (x^2 - 2x\xi + 2Ix y \sim + \xi^2 - 2I\xi y \sim - y \sim^2))$$

> sx1_in:=factor(evalc(Re(sx1_in)));evalc(Im(%));

$$sx1_in := \frac{4(x-\xi)^2 y \sim}{(x^2 - 2x\xi + \xi^2 + y \sim^2)^2}$$

0

Объединим в один интеграл и поменяем порядок интегрирования в интегралах с функцией $g(\xi)$ (предполагаем законной такую операцию)

```
> sx2:=1/2*I/Pi*Int(g(xi)*Int((y/s*abs(s)^2-
2/s*abs(s))*exp(-abs(s)*y+s*(x-xi)*I),s=-
infinity..infinity),xi=-infinity..infinity);
```

$$sx2 := \frac{1}{2} I \int_{-\infty}^{\infty} g(\xi) \int_{-\infty}^{\infty} \left(\frac{y|s|^2}{s} - \frac{2|s|}{s} \right) e^{(-|s|y+s(x-\xi)I)} ds d\xi$$

Выделяем внутренний интеграл и вычисляем его

```
> sx2_in:=Int(I*(y/s*abs(s)^2-2/s*abs(s))*exp(-
abs(s)*y+s*(x-xi)*I),s=-infinity..infinity);
```

$$sx2_in := \int_{-\infty}^{\infty} \left(\frac{y|s|^2}{s} - \frac{2|s|}{s} \right) e^{(-|s|y+s(x-\xi)I)} I ds$$

```
> simplify(evalc(Im(value(sx2_in))));
```

0

```
> sx2_in:=factor(simplify(evalc(Re(value(sx2_in))));
```

$$sx2_in := \frac{4(x-\xi)^3}{(x^2-2x\xi+\xi^2+y^2)^2}$$

Собираем теперь вместе общий внутренний интеграл

```
> sx_in:=f(xi)*sx1_in+g(xi)*sx2_in;
```

$$sx_in := \frac{4f(\xi)(x-\xi)^2 y}{(x^2-2x\xi+\xi^2+y^2)^2} + \frac{4g(\xi)(x-\xi)^3}{(x^2-2x\xi+\xi^2+y^2)^2}$$

Окончательно получаем формулу для напряжения σ_x

```
> sigma[x]:=1/2*1/Pi*Int(sx_in,xi=-infinity..infinity);
```

$$\sigma_x := \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \frac{4f(\xi)(x-\xi)^2 y}{(x^2-2x\xi+\xi^2+y^2)^2} + \frac{4g(\xi)(x-\xi)^3}{(x^2-2x\xi+\xi^2+y^2)^2} d\xi \right)$$

Таким образом, напряжение σ_x определяется по формуле

$$\sigma_x = \frac{2}{\pi} \int_{-\infty}^{+\infty} \frac{yf(\xi) + (x-\xi)g(\xi)}{[y^2 + (x-\xi)^2]^2} (x-\xi)^2 d\xi.$$

Преобразование формулы для напряжения σ_y .

```
> sigma['y'];
```

$$-\frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} e^{(-s|y)} G(s) s e^{(sxI)} I ds \right) + \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} e^{(-s|y)} F(s) e^{(sxI)} ds \right) + \frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} e^{(-s|y)} F(s) |s| e^{(sxI)} ds \right)$$

Подставим в эту формулу выражения $F(s)$ и $G(s)$ через интегралы

> $F(s) := \text{convert}(\text{fourier}(f(x_i), x_i, s), \text{int});$

$G(s) := \text{convert}(\text{fourier}(g(x_i), x_i, s), \text{int});$

$$F(s) := \int_{-\infty}^{\infty} f(\xi) e^{(-I\xi s)} d\xi$$

$$G(s) := \int_{-\infty}^{\infty} g(\xi) e^{(-I\xi s)} d\xi$$

Теперь напряжение σ_y имеет вид

> $\text{sigma}['y'];$

$$\begin{aligned} & -\frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} e^{(-s|y)} \int_{-\infty}^{\infty} g(\xi) e^{(-I\xi s)} d\xi s e^{(sxI)} I ds \right) \\ & + \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} e^{(-s|y)} \int_{-\infty}^{\infty} f(\xi) e^{(-I\xi s)} d\xi e^{(sxI)} ds \right) \\ & + \frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} e^{(-s|y)} \int_{-\infty}^{\infty} f(\xi) e^{(-I\xi s)} d\xi |s| e^{(sxI)} ds \right) \end{aligned}$$

%1 := $-\infty .. \infty$

Преобразуем отдельно интегралы, содержащие функцию $f(\xi)$ и интегралы, содержащие функцию $g(\xi)$. Объединим в один интеграл и поменяем порядок интегрирования в интегралах с функцией $f(\xi)$ (предполагаем законной такую операцию)

> $\text{sy1} := 1/2 * 1/\text{Pi} * \text{Int}(f(x_i) * \text{Int}(\exp(-\text{abs}(s) * y + s * (x - x_i) * I) * (1 + \text{abs}(s) * y), s = -\text{infinity} .. \text{infinity}), x_i = -\text{infinity} .. \text{infinity});$

$$syI := \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} f(\xi) \int_{-\infty}^{\infty} e^{(-s|y+s(x-\xi)I)} (1+|s|y) ds d\xi \right)$$

Выделяем внутренний интеграл и вычисляем его

```
> syI_in:=Int(exp(-abs(s)*y+s*(x-xi)*I)*(1+abs(s)*y),s=-infinity..infinity);
```

$$syI_in := \int_{-\infty}^{\infty} e^{(-s|y+s(x-\xi)I)} (1+|s|y) ds$$

```
> assume(y>0);
```

```
> syI_in:=simplify(value(syI_in));
```

$$syI_in := -4 y^3 / ((y^2 + 2 I x y - 2 I \xi y - x^2 + 2 x \xi - \xi^2) (x^2 - 2 x \xi + 2 I x y + \xi^2 - 2 I \xi y - y^2))$$

```
> syI_in:=factor(evalc(Re(syI_in)));
```

```
simplify(evalc(Im(%)));
```

$$syI_in := \frac{4 y^3}{(x^2 - 2 x \xi + \xi^2 + y^2)^2}$$

0

Поменяем порядок интегрирования в интеграле с функцией $g(\xi)$

```
> sy2:=-y/2/Pi*Int(g(xi)*Int(I*s*exp(-abs(s)*y+I*(x-xi)*s),s=-infinity..infinity),xi=-infinity..infinity);
```

$$sy2 := -\frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} g(\xi) \int_{-\infty}^{\infty} s e^{(-s|y+s(x-\xi)I)} I ds d\xi \right)$$

и упростим его

```
> sy2:=simplify(value(sy2));
```

$$sy2 := -\frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} 4 g(\xi) y (x - \xi) / ((y^2 + 2 I x y - 2 I \xi y - x^2 + 2 x \xi - \xi^2) (x^2 - 2 x \xi + 2 I x y + \xi^2 - 2 I \xi y - y^2)) d\xi \right)$$

Упростим подынтегральную функцию. Для этого выделяем ее

```
> student[integrand](sy2);
```

$$\frac{4 g(\xi) y (x - \xi)}{(y^2 + 2 I x y - 2 I \xi y - x^2 + 2 x \xi - \xi^2) (x^2 - 2 x \xi + 2 I x y + \xi^2 - 2 I \xi y - y^2)}$$

Выделяем отдельно знаменатель и упрощаем его

> expand(denom(%));

$$-y^4 + 4 x \xi^3 - 2 \xi^2 y^2 + 4 x \xi y^2 - 6 x^2 \xi^2 - 2 x^2 y^2 - \xi^4 - x^4 + 4 x^3 \xi$$

> factor(%);

$$-(x^2 - 2 x \xi + \xi^2 + y^2)^2$$

Итак, интеграл, содержащий функцию $g(\xi)$ преобразуется в следующий интеграл

> sy2:=2*y^2/Pi*Int(g(xi)*(x-xi)/(x^2-2*x*xi+xi^2+y^2)^2,xi=-infinity..infinity);

$$sy2 := \frac{2 y^2}{\pi} \int_{-\infty}^{\infty} \frac{g(\xi) (x - \xi)}{(x^2 - 2 x \xi + \xi^2 + y^2)^2} d\xi$$

Собираем теперь все вместе под общий интеграл и окончательно получаем формулу для напряжения σ_y

> sigma[y] := 2*y^2/Pi*Int((y*f(xi)+g(xi)*(x-xi))/(x^2-2*x*xi+xi^2+y^2)^2,xi = -infinity .. infinity);

$$\sigma_y := \frac{2 y^2}{\pi} \int_{-\infty}^{\infty} \frac{y f(\xi) + g(\xi) (x - \xi)}{(x^2 - 2 x \xi + \xi^2 + y^2)^2} d\xi$$

Итак, напряжение σ_y определяется по формуле

$$\sigma_y = \frac{2y^2}{\pi} \int_{-\infty}^{+\infty} \frac{yf(\xi) + (x - \xi)g(\xi)}{[y^2 + (x - \xi)^2]^2} d\xi.$$

Преобразование формулы для напряжения τ_{xy} .

> expand(tau);

$$\begin{aligned} & -\frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} \frac{|s| G(s) e^{(s x I)}}{e^{(|s| y)}} ds \right) + \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \frac{G(s) e^{(s x I)}}{e^{(|s| y)}} ds \right) \\ & + \frac{-1}{2} \frac{I y}{\pi} \int_{-\infty}^{\infty} \frac{|s|^2 F(s) e^{(s x I)}}{s e^{(|s| y)}} ds \end{aligned}$$

Подставим в последнюю формулу выражения $F(s)$ и $G(s)$ через интегралы

```
> F(s) := convert(fourier(f(xi), xi, s), int);
G(s) := convert(fourier(g(xi), xi, s), int);
```

$$F(s) := \int_{-\infty}^{\infty} f(\xi) e^{(-I\xi s)} d\xi$$

$$G(s) := \int_{-\infty}^{\infty} g(\xi) e^{(-I\xi s)} d\xi$$

Теперь напряжение τ_{xy} имеет вид

```
> expand(tau);
```

$$\begin{aligned} & \frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} \frac{|s| e^{(s x I)}}{e^{(|s| y)}} \int_{-\infty}^{\infty} \frac{g(\xi)}{e^{(\xi s I)}} d\xi ds \right) + \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \frac{e^{(s x I)}}{e^{(|s| y)}} \int_{-\infty}^{\infty} \frac{g(\xi)}{e^{(\xi s I)}} d\xi ds \right) \\ & \quad \quad \quad \%1 := -\infty .. \infty \\ & + \frac{-1}{2} \frac{I y}{\pi} \int_{-\infty}^{\infty} \frac{|s|^2 e^{(s x I)}}{s e^{(|s| y)}} \int_{-\infty}^{\infty} \frac{f(\xi)}{e^{(\xi s I)}} d\xi ds \end{aligned}$$

Преобразуем отдельно интегралы, содержащие функцию $f(\xi)$ и интегралы, содержащие функцию $g(\xi)$. Поменяем порядок интегрирования в интеграле с функцией $f(\xi)$ (предполагаем законной такую операцию)

```
> t1 := -I*y/2/Pi*Int(f(xi)*
Int(exp(-abs(s)*y+s*(x-xi)*I)*abs(s)^2/s,
s=-infinity..infinity), xi=-infinity..infinity);
```

$$t1 := \frac{-1}{2} \frac{I y}{\pi} \int_{-\infty}^{\infty} f(\xi) \int_{-\infty}^{\infty} \frac{e^{(-|s|y + s(x-\xi)I)} |s|^2}{s} ds d\xi$$

Выделяем внутренний интеграл и вычисляем его

```
> t1_in := Int(I*exp(-abs(s)*y+s*(x-xi)*I)*abs(s)^2/s,
s=-infinity..infinity);
```

$$t1_in := \int_{-\infty}^{\infty} \frac{e^{(-|s|y+s(x-\xi)I)} |s|^2 I}{s} ds$$

> assume (y>0) :

> t1_in:=simplify(value(t1_in));

$$t1_in := 4 y \sim (-x + \xi) / ((-y \sim^2 - 2 I x y \sim + 2 I \xi y \sim + x^2 - 2 x \xi + \xi^2) (x^2 - 2 x \xi + 2 I x y \sim + \xi^2 - 2 I \xi y \sim - y \sim^2))$$

> simplify(evalc(Re(t1_in))); simplify(evalc(Im(t1_in)));

$$\frac{4 y \sim (-x + \xi)}{x^4 - 4 x^3 \xi + 6 x^2 \xi^2 + 2 x^2 y \sim^2 - 4 x \xi^3 - 4 x \xi y \sim^2 + \xi^4 + 2 \xi^2 y \sim^2 + y \sim^4}$$

0

Упростим знаменатель дроби

> factor(denom(%%));

$$(\xi^2 + y \sim^2 - 2 x \xi + x^2)^2$$

Итак, интеграл, содержащий функцию $f(\xi)$, преобразовался к виду

> t1:=2*y/Pi*Int(y*f(xi)*(x-xi)/(x^2+y^2-2*x*xi+xi^2)^2, xi=-infinity..infinity);

$$t1 := \frac{2 y \sim}{\pi} \int_{-\infty}^{\infty} \frac{y \sim f(\xi) (x - \xi)}{(\xi^2 + y \sim^2 - 2 x \xi + x^2)^2} d\xi$$

Объединим в один интеграл и поменяем порядок интегрирования в интегралах с функцией $g(\xi)$:

> t2:=1/2/Pi*Int(g(xi)*

Int((1-y*abs(s))*exp(-abs(s)*y+s*(x-xi)*I),

s=-infinity..infinity), xi=-infinity..infinity);

$$t2 := \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} g(\xi) \int_{-\infty}^{\infty} (1 - |s| y \sim) e^{(-|s| y \sim + s(x - \xi) I)} ds d\xi \right)$$

$$-\frac{1}{2} \left(\frac{y}{\pi} \int_{-\infty}^{\infty} \frac{|s| e^{(s x I)}}{e^{(|s| y)}} \int_{-\infty}^{\infty} \frac{g(\xi)}{e^{(\xi s I)}} d\xi ds \right) + \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \frac{e^{(s x I)}}{e^{(|s| y)}} \int_{-\infty}^{\infty} \frac{g(\xi)}{e^{(\xi s I)}} d\xi ds \right)$$

Выделяем внутренний интеграл и вычисляем его

```
> t2_in:=Int((1-y*abs(s))*exp(-abs(s)*y+s*(x-xi)*I),
s=-infinity..infinity);
```

$$t2_in := \int_{-\infty}^{\infty} (1 - |s| y) e^{(-|s| y + s(x - \xi) I)} ds$$

```
> simplify(evalc(Im(value(t2_in))));
0
```

```
> t2_in:=factor(simplify(evalc(Re(value(t2_in))));
```

$$t2_in := \frac{4(-x + \xi)^2 y}{(\xi^2 + y^2 - 2x\xi + x^2)^2}$$

Таким образом, интеграл, содержащий функцию $g(\xi)$, преобразовался к виду

```
> t2:=2*y/Pi*Int(g(xi)*(x-xi)^2/(x^2+y^2-2*x*xi+xi^2)^2,
xi=-infinity..infinity);
```

$$t2 := \frac{2y}{\pi} \int_{-\infty}^{\infty} \frac{g(\xi)(x - \xi)^2}{(\xi^2 + y^2 - 2x\xi + x^2)^2} d\xi$$

Окончательно получаем формулу для напряжения τ_{xy} :

```
> 'tau[xy]'=2*y/Pi*Int(student[integrand](t1)+
student[integrand](t2),xi=-infinity..infinity);
```

$$\tau_{xy} = \frac{2y}{\pi} \int_{-\infty}^{\infty} \frac{y f(\xi)(x - \xi)}{(\xi^2 + y^2 - 2x\xi + x^2)^2} + \frac{g(\xi)(x - \xi)^2}{(\xi^2 + y^2 - 2x\xi + x^2)^2} d\xi$$

Таким образом, напряжение τ_{xy} определяется по формуле

$$\tau_{xy} = \frac{2y}{\pi} \int_{-\infty}^{+\infty} \frac{y f(\xi) + (x - \xi) g(\xi)}{[y^2 + (x - \xi)^2]^2} (x - \xi) d\xi.$$

Задача 6.2. С помощью преобразования Фурье найти решение уравнения

$$\frac{1}{a^2} \frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}$$

в области $-\infty < x < +\infty$, $0 \leq t < +\infty$, при начальных условиях

$$u|_{t=0} = \varphi(x), \quad \left. \frac{\partial u}{\partial t} \right|_{t=0} = \psi(x).$$

Решение. Определим задачу в Maple

```
> restart;
> with(inttrans);
[ addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert,
  invlaplace, invmellin, laplace, mellin, savetable ]
> PDE:=1/a^2*diff(u(x,t),t$2)=diff(u(x,t),x$2);
init_c1:=u(x,0)=phi(x);
init_c2:=D[2](u)(x,0)=psi(x);
```

$$PDE := \frac{\frac{\partial^2}{\partial t^2} u(x, t)}{a^2} = \frac{\partial^2}{\partial x^2} u(x, t)$$

$$init_c1 := u(x, 0) = \phi(x)$$

$$init_c2 := D_2(u)(x, 0) = \psi(x)$$

Применяем преобразование Фурье к заданному уравнению

```
> de:=fourier(PDE,x,s);
```

$$de := \frac{\frac{\partial^2}{\partial t^2} \text{fourier}(u(x, t), x, s)}{a^2} = -s^2 \text{fourier}(u(x, t), x, s)$$

Введем обозначение преобразования Фурье $\text{fourier}(u(x, t), x, s)$ как $U(t)$

```
> de:=subs(fourier(u(x,t),x,s)=U(t),de);
```

$$de := \frac{d^2}{dt^2} U(t) = -s^2 U(t)$$

Применяем преобразование Фурье к начальным условиям

```
> f_init1:=fourier(init_c1,x,s);
f_init1:=subs(fourier(u(x,0),x,s)=U(0),
fourier(phi(x),x,s)=Phi,f_init1);
```

```

f_init2:=fourier(init_c2,x,s);
f_init2:=subs(fourier(D[2](u)(x,0),x,s)=D(U)(0),
fourier(psi(x),x,s)=Psi,f_init2);
f_init1 := fourier(u(x,0),x,s) = fourier(phi(x),x,s)
f_init1 := U(0) = Phi
f_init2 := fourier(D_2(u)(x,0),x,s) = fourier(psi(x),x,s)
f_init2 := D(U)(0) = Psi

```

Решаем теперь полученное обыкновенное дифференциальное уравнение для трансформанты Фурье с соответствующими начальными условиями

```
> dsolve({de,f_init1,f_init2},U(t));
```

$$U(t) = \frac{\Psi \sin(a s t)}{a s} + \Phi \cos(a s t)$$

Находим теперь обратное преобразование Фурье

```
> subs(U(t)=fourier(u(x,t),x,s),Phi=Phi(s),Psi=Psi(s),%);
```

$$\text{fourier}(u(x,t),x,s) = \frac{\Psi(s) \sin(a s t)}{a s} + \Phi(s) \cos(a s t)$$

```
> res:=expand(convert(invfourier(%,s,x),int));
```

```
res := u(x,t) =
```

$$\frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \Phi(s) \cos(a s t) e^{(s x t)} ds \right) + \frac{1}{2} \left(\frac{1}{a \pi} \int_{-\infty}^{\infty} \frac{\Psi(s) \sin(a s t) e^{(s x t)}}{s} ds \right)$$

Итак, решение задачи дается формулой

```
> assign(res):u:=unapply(u(x,t),x,t);
```

```
u := (x,t) →
```

$$\frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \Phi(s) \cos(a s t) e^{(s x t)} ds \right) + \frac{1}{2} \left(\frac{1}{a \pi} \int_{-\infty}^{\infty} \frac{\Psi(s) \sin(a s t) e^{(s x t)}}{s} ds \right)$$

где

```
> Psi(s)=convert(fourier(psi(x),x,s),int);
```

```
Phi(s)=convert(fourier(phi(x),x,s),int);
```

$$\Psi(s) = \int_{-\infty}^{\infty} \psi(x) e^{(-I x s)} dx$$

$$\Phi(s) = \int_{-\infty}^{\infty} \phi(x) e^{(-Ix s)} dx$$

Сделаем проверку полученного решения

> simplify (expand (rhs (PDE) - lhs (PDE))) ;

0

> simplify (init_c1) ; simplify (init_c2) ;

$$\frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \Phi(s) e^{(s x I)} ds \right) = \phi(x)$$

$$\frac{1}{2} \left(\frac{1}{a \pi} \int_{-\infty}^{\infty} \Psi(s) a e^{(s x I)} ds \right) = \psi(x)$$

Видим, что уравнение тождественно выполняется, а последние две формулы есть ни что иное, как формулы обращения преобразования Фурье для начальных функций, то есть и начальные условия выполняются.

Таким образом, мы получили решение задачи в виде

$$u(x, t) = \frac{1}{2\pi} \int_{-\infty}^{+\infty} \Phi(s) e^{isx} \cos(ast) ds + \frac{1}{2a\pi} \int_{-\infty}^{+\infty} \frac{\Psi(s) e^{isx} \sin(ast)}{s} ds,$$

где

$$\Phi(s) = \int_{-\infty}^{+\infty} \phi(x) e^{-isx} dx, \quad \Psi(s) = \int_{-\infty}^{+\infty} \psi(x) e^{-isx} dx.$$

Как мы знаем, решение поставленной задачи дается формулой Даламбера (см. глава 3, формула (3.17)). Покажем, как преобразовать полученное решение в формулу Даламбера. Для этого докажем справедливость следующих формул

$$\frac{1}{2\pi} \int_{-\infty}^{+\infty} \frac{F(s) e^{isx} \sin(st)}{s} ds = \frac{1}{2} \int_{x-t}^{x+t} f(\xi) d\xi, \quad (*)$$

$$\frac{1}{2\pi} \int_{-\infty}^{+\infty} F(s) e^{isx} \cos(ast) ds = \frac{f(x+t) + f(x-t)}{2}, \quad (**)$$

где $F(s)$ преобразование Фурье функции $f(x)$, то есть

$$\frac{1}{2\pi} \int_{-\infty}^{+\infty} F(s) e^{isx} = f(x), \quad (***)$$

$$\int_{-\infty}^{+\infty} f(x) e^{-isx} ds = F(s).$$

Заметим, что формула (***) получается из формулы (*) дифференцирование по параметру t , а поэтому достаточно доказать формулу (*). Действительно, проверим это утверждение — определим формулу (*) в Maple и продифференцируем ее

```
> 1/2/Pi*Int (exp (I*s*x) *F(s) *sin(s*t) /s,
s=-infinity..infinity) =
```

```
1/2*Int (f (xi) , xi=x-t..x+t) ;
```

$$\frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \frac{e^{(s x I)} F(s) \sin(s t)}{s} ds \right) = \frac{1}{2} \int_{x-t}^{x+t} f(\xi) d\xi$$

```
> diff (% , t) ;
```

$$\frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} e^{(s x I)} F(s) \cos(s t) ds \right) = \frac{1}{2} f(x+t) + \frac{1}{2} f(x-t)$$

В результате получили формулу (**).

Докажем формулу (*). Для этого заменим в ней синус по формуле Эйлера

```
> p1 := combine (expand (convert (lhs (%%) , exp))) ;
```

$$p1 := \int_{-\infty}^{\infty} \frac{-1}{4} I \left(\frac{F(s) e^{(s x I + s t I)}}{s} - \frac{F(s) e^{(s x I - s t I)}}{s} \right) ds$$

Таким образом, мы получили

$$\begin{aligned} & \frac{1}{2\pi} \int_{-\infty}^{+\infty} \frac{F(s) e^{isx} \sin(st)}{s} ds = \\ & = -\frac{i}{4\pi} \int_{-\infty}^{+\infty} \frac{F(s) e^{is(x+t)}}{s} ds + \frac{i}{4\pi} \int_{-\infty}^{+\infty} \frac{F(s) e^{is(x-t)}}{s} ds = \\ & = \frac{g(x+t) - g(x-t)}{2}, \end{aligned}$$

где введено обозначение

$$g(x) = -\frac{i}{2\pi} \int_{-\infty}^{+\infty} \frac{F(s) e^{isx}}{s} ds.$$

Действительно, проверим это с помощью Maple — определим функцию $g(x)$

```
> g := (x) -> 1/2/Pi*Int (exp (I*s*x) *F (s) / (I*s) ,
s=-infinity..infinity) ;
```

$$g := x \rightarrow \frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} \frac{-I e^{(s x I)} F(s)}{s} ds \right)$$

Вычислим правую часть формулы, то есть выражение $[g(x+t) - g(x-t)]/2$, и сравним это выражение с левой частью

```
> p2 := combine (1/2* (g (x+t) -g (x-t) ) ) ;
simplify (p1-p2) ;
```

$$p2 := \int_{-\infty}^{\infty} -\frac{\frac{1}{4} I e^{(s(x+t)I)} F(s)}{\pi s} + \frac{\frac{1}{4} I e^{(s(x-t)I)} F(s)}{\pi s} ds$$

0

Как видим, они совпадают.

Продифференцируем функцию $g(x)$

```
> diff (g (x) , x) ;
```

$$\frac{1}{2} \left(\frac{1}{\pi} \int_{-\infty}^{\infty} e^{(s x I)} F(s) ds \right)$$

Последний интеграл есть обращение преобразования Фурье, то есть функция $f(x)$ (см. формулу (***)). Таким образом, можно записать

$$\frac{dg}{dx} = f(x) \Rightarrow g(x) = \int_A^x f(\xi) d\xi - g(A)$$

и

$$g(x+t) - g(x-t) = \int_A^{x+t} f(\xi) d\xi - \int_A^{x-t} f(\xi) d\xi = \int_{x-t}^{x+t} f(\xi) d\xi,$$

что и доказывает формулу (*).

Таким образом, решение задачи можно переписать в виде формулы Даламбера

$$u(x,t) = \frac{\varphi(x+at) + \varphi(x-at)}{2} + \frac{1}{2a} \int_{x-at}^{x+at} \psi(\xi) d\xi.$$

Задача 6.3. С помощью преобразования Фурье найти решение уравнения

$$\frac{1}{a^2} \frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2}$$

в области $-\infty < x < +\infty$, $0 \leq t < +\infty$, при начальных условиях

$$u|_{t=0} = \varphi(x).$$

Решение. Определим задачу в Maple

```
> restart;
> with(inttrans):
> deq:=1/a^2*difff(u(x,t),t)=difff(u(x,t),x$2);
```

$$deq := \frac{\frac{\partial}{\partial t} u(x, t)}{a^2} = \frac{\partial^2}{\partial x^2} u(x, t)$$

```
> ic:=u(x,0)=phi(x);
```

$$ic := u(x, 0) = \varphi(x)$$

Применяем преобразование Фурье к исходному уравнению в частных производных

```
> feq:=fourier(deq,x,s);
```

$$feq := \frac{\frac{\partial}{\partial t} \text{fourier}(u(x, t), x, s)}{a^2} = -s^2 \text{fourier}(u(x, t), x, s)$$

```
> feq:=subs(fourier(u(x,t),x,s)=U(t),feq);
```

$$feq := \frac{d}{dt} U(t) = -s^2 U(t)$$

Здесь, как и в предыдущей задаче, трансформанту Фурье мы обозначили как $U(t)$.

Трансформируем теперь начальное условие

```
> icf:=fourier(ic,x,s);
```

```
icf:=subs(fourier(u(x,0),x,s)=U(0),
```

```
fourier(phi(x),x,s)=Phi,icf);
```

$$icf := \text{fourier}(u(x, 0), x, s) = \text{fourier}(\varphi(x), x, s)$$

$$icf := U(0) = \Phi$$

Здесь через Φ обозначена трансформанта Фурье функции φ .

Решаем уравнение для трансформанты с учетом соответствующего начального условия

> dsolve({feq, icf}, U(t));

$$U(t) = \Phi e^{(-s^2 a^2 t)}$$

Находим обратное преобразование

> assign(%):subs(Phi=fourier(phi(x), x, s), U(t));

sol:=invsfourier(% , s, x);

$$\text{fourier}(\phi(x), x, s) e^{(-s^2 a^2 t)}$$

$$\text{sol} := \text{invsfourier}(\text{fourier}(\phi(x), x, s) e^{(-s^2 a^2 t)}, s, x)$$

Итак, решение задачи дается формулой

> assume(t>0):assume(a>0):convert(sol, int);

$$\frac{1}{4} \left(\frac{1}{\pi^2} \int_{-\infty}^{\infty} \frac{2 \pi^{(3/2)} \phi(-U) e^{\left(-\frac{(x+U)^2}{4 a^2 t}\right)}}{a \sqrt{t}} d_U \right)$$

Преобразуем это выражение к общепринятым обозначениям и одновременно упростим его

> student[changevar](_U1=-xi, %, xi);

$$\frac{1}{4} \left(\frac{1}{\pi^2} \int_{-\infty}^{\infty} \frac{2 \pi^{(3/2)} \phi(\xi) e^{\left(-\frac{(x-\xi)^2}{4 a^2 t}\right)}}{a \sqrt{t}} d\xi \right)$$

> simplify(expand(%));

$$\frac{1}{2} \left(\frac{1}{\sqrt{\pi} a \sqrt{t}} \int_{-\infty}^{\infty} \phi(\xi) e^{\left(-\frac{(x-\xi)^2}{4 a^2 t}\right)} d\xi \right)$$

Итак, мы получили решение задачи в виде интеграла

> u:=unapply(% , x, t);

$$u := (x, t) \rightarrow \frac{1}{2} \left(\frac{1}{\sqrt{\pi} a \sqrt{t}} \int_{-\infty}^{\infty} \phi(\xi) e^{\left(-1/4 \frac{(x-\xi)^2}{a^2 t}\right)} d\xi \right)$$

> 'u(x, t)'=u(x, t);

$$u(x, t) = \frac{1}{2} \left(\frac{1}{\sqrt{\pi a \sqrt{t}}} \int_{-\infty}^{\infty} \phi(\xi) e^{-\frac{(x-\xi)^2}{4a^2 t}} d\xi \right)$$

Выпишем окончательную формулу решения нашей задачи

$$u(x, t) = \frac{1}{2a\sqrt{\pi t}} \int_{-\infty}^{+\infty} \phi(\xi) e^{-\frac{(x-\xi)^2}{4a^2 t}} d\xi.$$

Задача 6.4. Найти стационарное распределение температуры в полубесконечном теле ($x \geq 0, y \geq 0$), если одна грань ($x = 0$) излучает тепло по закону Ньютона (температура среды принимается равной нулю), а другая грань ($y = 0$) поддерживается при постоянной температуре T_0 .

Решение. Сформулируем математическую задачу: найти функцию $u(x, t)$, удовлетворяющую стационарному уравнению теплопроводности

$$\Delta u \equiv \frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} = 0$$

и граничным условиям

$$u|_{y=0} = T_0, \quad \frac{\partial u}{\partial x} - hu \Big|_{x=0} = 0.$$

Определим уравнение в Maple

```
> deq:=linalg[laplacian](u(x,y),[x,y])=0;
```

$$deq := \left(\frac{\partial^2}{\partial x^2} u(x, y) \right) + \left(\frac{\partial^2}{\partial y^2} u(x, y) \right) = 0$$

Применяем к заданному уравнению в частных производных синус-преобразование Фурье по переменной y

```
> feq:=inttrans[fouriersin](lhs(deq),y,s)=0;
```

$$feq := \left(\frac{\partial^2}{\partial x^2} \text{fouriersin}(u(x, y), y, s) \right) - \frac{s(-\sqrt{2} u(x, 0) + s \text{fouriersin}(u(x, y), y, s) \sqrt{\pi})}{\sqrt{\pi}} = 0$$

```
> feq:=subs(fouriersin(u(x,y),y,s)=U(x),u(x,0)=T[0],feq);
```

$$feq := \left(\frac{d^2}{dx^2} U(x) \right) - \frac{s(-\sqrt{2} T_0 + s U(x) \sqrt{\pi})}{\sqrt{\pi}} = 0$$

Здесь трансформанту Фурье мы обозначили как $U(x)$.

Решаем уравнение для трансформанты

> dsolve (feq, U (x));

$$U(x) = e^{(-s x)} _C2 + e^{(s x)} _C1 + \frac{\sqrt{2} T_0}{s \sqrt{\pi}}$$

В силу ограниченности функции на бесконечности полагаем $_C1=0$ ($s > 0, x > 0$)

> assign (%) :subs (_C1=0, U (x));

$$e^{(-s x)} _C2 + \frac{\sqrt{2} T_0}{s \sqrt{\pi}}$$

> U:=unapply (% , x) ;

$$U := x \rightarrow e^{(-s x)} _C2 + \frac{\sqrt{2} T_0}{s \sqrt{\pi}}$$

Находим вторую константу из условия по переменной x

> eq:=D (U) (0) -h*U (0) =0;

$$eq := -s _C2 - h \left(_C2 + \frac{\sqrt{2} T_0}{s \sqrt{\pi}} \right) = 0$$

> _C2:=solve (eq, _C2) ;

$$_C2 := -\frac{h \sqrt{2} T_0}{s \sqrt{\pi} (s + h)}$$

Итак, синус-трансформанта Фурье найдена

> U (x) ;

$$-\frac{e^{(-s x)} h \sqrt{2} T_0}{s \sqrt{\pi} (s + h)} + \frac{\sqrt{2} T_0}{s \sqrt{\pi}}$$

Находим теперь обратное преобразование Фурье

> assume (x>0) :

> inttrans [fouriersin] (% , s, y) ;

$$-h \sqrt{2} T_0 \text{fouriersin} \left(\frac{e^{(-s x)}}{s \sqrt{\pi} (s + h)}, s, y \right) + T_0$$

> convert (% , Int) ;

$$\left[\frac{2 h T_0}{\sqrt{\pi}} \int_0^{\infty} \frac{e^{(-s x)} \sin(s y)}{s \sqrt{\pi} (s + h)} ds \right] + T_0$$

Таким образом, решение задачи дается следующим интегралом

$$u(x, t) = T_0 \left[1 - \frac{2h}{\pi} \int_0^{\infty} \frac{e^{-sx} \sin(sy)}{s(s+h)} ds \right],$$

который оформим в Maple как функцию

> u := unapply(%, x, y);

$$u := (x~, y) \rightarrow - \left(\frac{2 h T_0}{\sqrt{\pi}} \int_0^{\infty} \frac{e^{(-s x~)} \sin(s y)}{s \sqrt{\pi} (s + h)} ds \right) + T_0$$

Выполним проверку решения. Проверяем уравнение

> deq;

$$- \left(\frac{2 h T_0}{\sqrt{\pi}} \int_0^{\infty} \frac{s e^{(-s x~)} \sin(s y)}{\sqrt{\pi} (s + h)} ds \right) - \left(\frac{2 h T_0}{\sqrt{\pi}} \int_0^{\infty} - \frac{s e^{(-s x~)} \sin(s y)}{\sqrt{\pi} (s + h)} ds \right) = 0$$

> expand(deq);

$$0 = 0$$

Уравнение удовлетворяется тождественно.

Проверяем граничные условия

> value(u(x, 0));

$$T_0$$

> assume(y>0): D[1](u)(0, y) - h*u(0, y);

$$- \left(\frac{2 h T_0}{\sqrt{\pi}} \int_0^{\infty} - \frac{\sin(s y~)}{\sqrt{\pi} (s + h)} ds \right) - h \left(- \left(\frac{2 h T_0}{\sqrt{\pi}} \int_0^{\infty} \frac{\sin(s y~)}{s \sqrt{\pi} (s + h)} ds \right) + T_0 \right)$$

> expand(value(%));

$$0$$

Все в порядке.

Задача 6.5. Решить задачу Коши для уравнения теплопроводности

$$\frac{\partial u}{\partial t} = 4 \frac{\partial^2 u}{\partial x^2} + t + e^t$$

в области $-\infty < x < +\infty$, $0 \leq t < +\infty$, при начальных условиях

$$u|_{t=0} = 2.$$

Решение. Для решения задачи применим преобразование Фурье. Выполним все вычисления в Maple

```

> restart;
> with(inttrans);
[ addtable , fourier , fouriercos , fouriersin , hankel , hilbert , invfourier , invhilbert ,
  invlaplace , invmellin , laplace , mellin , savetable ]
> deq:=diff(u(x,t),t)=4*diff(u(x,t),x$2)+t+exp(t);

$$deq := \frac{\partial}{\partial t} u(x, t) = 4 \left( \frac{\partial^2}{\partial x^2} u(x, t) \right) + t + e^t$$

> feq:=fourier(deq,x,s);

$$feq := \frac{\partial}{\partial t} \text{fourier}(u(x, t), x, s) =$$


$$-4 s^2 \text{fourier}(u(x, t), x, s) + 2 e^t \pi \text{Dirac}(s) + 2 t \pi \text{Dirac}(s)$$

> feq:=subs(fourier(u(x,t),x,s)=U(t),feq);

$$feq := \frac{d}{dt} U(t) = -4 s^2 U(t) + 2 e^t \pi \text{Dirac}(s) + 2 t \pi \text{Dirac}(s)$$

> bond:=fourier(u(x,0)=2,x,s);
bond:=subs(fourier(u(x,0),x,s)=U(0),bond);

$$bond := \text{fourier}(u(x, 0), x, s) = 4 \pi \text{Dirac}(s)$$


$$bond := U(0) = 4 \pi \text{Dirac}(s)$$

> dsolve({feq,bond},U(t));

$$U(t) = (\pi \text{Dirac}(s) (2 e^t + t^2) + 2 \pi \text{Dirac}(s)) e^{(-4 s^2 t)}$$


```

Итак, мы получили трансформанту Фурье в виде

$$U(t) = \pi e^{-4s^2 t} \left[(2e^t + t^2) + 2 \right] \delta(s),$$

где $\delta(s)$ — дельта-функция Дирака.

Находим теперь обратное преобразование

```

> assign(%) :U(t);
invfourier(%,s,x);

$$(\pi \text{Dirac}(s) (2 e^t + t^2) + 2 \pi \text{Dirac}(s)) e^{(-4 s^2 t)}$$


$$e^t + \frac{t^2}{2} + 1$$


```

Таким образом, решение задачи имеет вид

$$u = e^t + \frac{t^2}{2} + 1.$$

Выполним проверку решения. Для этого определяем функцию

```
> u:=unapply(% , x, t);
```

$$u := (x, t) \rightarrow e^t + \frac{1}{2} t^2 + 1$$

Проверяем уравнение

```
> deq;
```

$$e^t + t = e^t + t$$

Проверяем начальное условие

```
> u(x, 0);
```

2

Все в порядке!

Задача 6.6. Найти распределение потенциала в цилиндрической электронной линзе², изображенной на рис. 6.2




Рис. 6.2. Цилиндрическая электронная линза

Решение. Сформулируем математическую задачу: найти функцию $u(r, z)$, удовлетворяющую уравнению Лапласа и дополнительным условиям

$$u|_{z \rightarrow \infty} < \infty, \quad u|_{r \rightarrow 0} < \infty, \quad u|_{r=a} = \psi(z) = \begin{cases} V_1, & z < 0, \\ V_2, & z > 0. \end{cases}$$

Сведем рассматриваемую задачу к задаче с граничными условиями, нечетными по переменной z . Для этого положим

$$u(r, z) = \frac{V_1 + V_2}{2} + v(r, z). \quad (*)$$

Выполним все вычисления в Maple. Задаем уравнение и дополнительные условия

```
> restart;with(linalg):with(student):
```

² Под электронооптическими устройствами (линзами и т.д.) понимаются системы проводников при заданных потенциалах, создающие электростатические поля, предназначенные служить для направления пучков заряженных частиц.

Warning, the protected names norm and trace have been redefined and unprotected

> pde:=laplacian(u(r,z),[r,phi,z], coords=cylindrical);

$$pde := \frac{\left(\frac{\partial}{\partial r} u(r,z)\right) + r \left(\frac{\partial^2}{\partial r^2} u(r,z)\right) + r \left(\frac{\partial^2}{\partial z^2} u(r,z)\right)}{r}$$

> bc:=u(a,z)=psi(z);

$$bc := u(a,z) = \psi(z)$$

> psi:=z->piecewise(z<0,V[1],z>=0,V[2]);psi(z);

$$\psi := z \rightarrow \text{piecewise}(z < 0, V_1, 0 \leq z, V_2)$$

$$\begin{cases} V_1 & z < 0 \\ V_2 & 0 \leq z \end{cases}$$

Определяем функцию $u(r,z)$ в соответствии с формулой (*)

> u:=(r,z)->(V[1]+V[2])/2+v(r,z);

$$u := (r,z) \rightarrow \frac{1}{2} V_1 + \frac{1}{2} V_2 + v(r,z)$$

Тогда исходное уравнение и граничные условия преобразовываются к виду

> pdel:=pde;

$$pdel := \frac{\left(\frac{\partial}{\partial r} v(r,z)\right) + r \left(\frac{\partial^2}{\partial r^2} v(r,z)\right) + r \left(\frac{\partial^2}{\partial z^2} v(r,z)\right)}{r}$$

> bc;isolate(%,v(a,z));

$$\frac{1}{2} V_1 + \frac{1}{2} V_2 + v(a,z) = \begin{cases} V_1 & z < 0 \\ V_2 & 0 \leq z \end{cases}$$

$$v(a,z) = \left(\begin{cases} V_1 & z < 0 \\ V_2 & 0 \leq z \end{cases} \right) - \frac{1}{2} V_1 - \frac{1}{2} V_2$$

> bc1:=simplify(%);

$$bc1 := v(a,z) = \begin{cases} \frac{1}{2} V_1 - \frac{1}{2} V_2 & z < 0 \\ -\frac{1}{2} V_1 + \frac{1}{2} V_2 & 0 \leq z \end{cases}$$

Итак, мы пришли к задаче:

$$\frac{\partial^2 v}{\partial r^2} + \frac{1}{r} \frac{\partial v}{\partial r} + \frac{\partial^2 v}{\partial z^2} = 0$$

$$v|_{z \rightarrow \infty} < \infty, \quad v|_{r \rightarrow 0} < \infty, \quad v|_{r=a} = p(z) = \begin{cases} \frac{V_1 - V_2}{2}, & z < 0, \\ \frac{V_2 - V_1}{2}, & z > 0. \end{cases}$$

Функцию $p(z)$ можно записать в виде

$$p(z) = \frac{V_2 - V_1}{2} \operatorname{signum}(z),$$

где $\operatorname{signum}(z) = \frac{z}{|z|}$ — знак числа z .

То есть

$$p := z \rightarrow (V[2] - V[1]) / 2 * \operatorname{signum}(z);$$

$$p := z \rightarrow \frac{1}{2} (V_2 - V_1) \operatorname{signum}(z)$$

Выполним теперь разделение переменных и сформулируем соответствующую задачу Штурма-Лиувилля

$$> \text{res} := \text{pdsolve}(\text{pdel}, \text{HINT} = \text{R}(r) * \text{Z}(z));$$

$$\text{res} := (v(r, z) = \text{R}(r) \text{Z}(z)) \&\text{where}$$

$$\left[\left\{ \frac{d^2}{dr^2} \text{R}(r) = \text{R}(r) _c_1 - \frac{d}{dr} \text{R}(r), \frac{d^2}{dz^2} \text{Z}(z) = -\text{Z}(z) _c_1 \right\} \right]$$

$$> \text{ode1} := \text{op}(1, \text{op}(1, \text{op}(2, \text{res})));$$

$$\text{ode2} := \text{op}(2, \text{op}(1, \text{op}(2, \text{res})));$$

$$\text{ode1} := \frac{d^2}{dr^2} \text{R}(r) = \text{R}(r) _c_1 - \frac{d}{dr} \text{R}(r)$$

$$\text{ode2} := \frac{d^2}{dz^2} \text{Z}(z) = -\text{Z}(z) _c_1$$

$$> \text{ode1} := \text{subs}(_c[1] = \text{lambda}, \text{ode1});$$

$$\text{ode2} := \text{subs}(_c[1] = \text{lambda}, \text{ode2});$$

$$\text{ode1} := \frac{d^2}{dr^2} \text{R}(r) = \text{R}(r) \lambda - \frac{d}{dr} \text{R}(r)$$

$$\text{ode2} := \frac{d^2}{dz^2} \text{Z}(z) = -\text{Z}(z) \lambda$$

$$> \text{ode1} := \text{collect}(\text{DEtools}[\text{DENormal}](\text{ode1}, r, \text{R}(r)), \text{R}(r));$$

```
ode2:=collect(DEtools[DEnormal](ode2,z,Z(z)),Z(z));
```

$$ode1 := -r \lambda R(r) + \left(\frac{d}{dr} R(r) \right) + \left(\frac{d^2}{dr^2} R(r) \right) r = 0$$

$$ode2 := \left(\frac{d^2}{dz^2} Z(z) \right) + Z(z) \lambda = 0$$

Таким образом, задача Штурма-Лиувилля имеет вид

$$Z''(z) + \lambda Z(z) = 0, \quad Z|_{z \rightarrow \infty} < \infty.$$

Находим общее решение уравнения Штурма-Лиувилля

```
> dsolve(ode2, Z(z));
```

$$Z(z) = _C1 \sin(\sqrt{\lambda} z) + _C2 \cos(\sqrt{\lambda} z)$$

Решение будет ограниченным на бесконечности, если $\lambda > 0$. Таким образом, мы имеем непрерывный спектр собственных значений $\lambda = \lambda_\nu = \nu^2$, $0 < \nu < +\infty$. Соответствующие собственные функции будут

$$Z = Z_\nu(z) = A_\nu \cos(\nu z) + B_\nu \sin(\nu z).$$

Или, в Maple

```
> Z := (nu, z) -> A(nu) * cos(nu*z) + B(nu) * sin(nu*z);
```

$$Z := (\nu, z) \rightarrow A(\nu) \cos(\nu z) + B(\nu) \sin(\nu z)$$

Решаем второе уравнение

```
> dsolve(subs(lambda=nu^2,ode1),R(r));
```

$$R(r) = _C1 \text{BesselI}(0, \nu r) + _C2 \text{BesselK}(0, \nu r)$$

В силу ограниченности искомой функции в нуле мы должны положить константу $_C2$ равной нулю, то есть будем иметь

$$R_\nu(r) = I_0(\nu r)$$

Или, в Maple

```
> R := (nu, r) -> BesselI(0, nu*r);
```

$$R := (\nu, r) \rightarrow \text{BesselI}(0, \nu r)$$

Таким образом, в силу обобщенного принципа суперпозиции имеем следующее разложение

$$v(r, z) = \int_0^\infty [A_\nu \cos(\nu z) + B_\nu \sin(\nu z)] I_0(\nu r) d\nu$$

Или в Maple

```
> v := Int(Z(nu, z) * R(nu, r), nu=0..infinity);
```

$$v := \int_0^{\infty} (A(v) \cos(vz) + B(v) \sin(vz)) \text{BesselI}(0, va) dv$$

Для определения коэффициентов разложения воспользуемся граничным условием

$$p(z) = \int_0^{\infty} [A_v \cos(vz) + B_v \sin(vz)] I_0(va) dv$$

Или в Maple

```
> 'p(z)' = subs(r=a, v);
```

$$p(z) = \int_0^{\infty} (A(v) \cos(vz) + B(v) \sin(vz)) \text{BesselI}(0, va) dv$$

Откуда находим

$$A_v I_0(va) = \frac{1}{\pi} \int_0^{\infty} p(z) \cos(vz) dz,$$

$$B_v I_0(va) = \frac{1}{\pi} \int_0^{\infty} p(z) \sin(vz) dz$$

Или в Maple

```
> eq1 := A(nu) * BesselI(0, nu*a) =
1/Pi * Int('p(z)' * cos(nu*z), z=-infinity..infinity);
eq2 := B(nu) * BesselI(0, nu*a) =
1/Pi * Int('p(z)' * sin(nu*z), z=-infinity..infinity);
```

$$eq1 := A(v) \text{BesselI}(0, va) = \frac{1}{\pi} \int_{-\infty}^{\infty} p(z) \cos(vz) dz$$

$$eq2 := B(v) \text{BesselI}(0, va) = \frac{1}{\pi} \int_{-\infty}^{\infty} p(z) \sin(vz) dz$$

Вычисляем интеграл справа в первом соотношении

```
> p1 :=
1/Pi * Int((V[1]-V[2])/2*cos(nu*z), z=-
infinity..0) + 1/Pi * Int((V[2]-
V[1])/2*cos(nu*z), z=0..infinity);
```

$$p1 := \frac{1}{\pi} \int_{-\infty}^0 \frac{1}{2} (V_1 - V_2) \cos(vz) dz + \frac{1}{\pi} \int_0^{\infty} \frac{1}{2} (V_2 - V_1) \cos(vz) dz$$

```
> changevar(z=-xi, op(1, p1), xi);
```

$$\frac{1}{\pi} \int_0^{\infty} \frac{1}{2} (V_1 - V_2) \cos(v \xi) d\xi$$

> p1:=expand(subs(op(1,p1)=%,xi=z,p1));
p1:=0

> A(nu):=0;

$$A(v) := 0$$

Таким образом, $A_v = 0, \forall v > 0$.

Следовательно, мы имеем разложение только по синусам

$$v(r, z) = \int_0^{\infty} B_v \sin(vz) I_0(vr) dv$$

или

> 'v(r, z)'=v;

$$v(r, z) = \int_0^{\infty} B(v) \sin(vz) \text{Bessel}(0, vr) dv$$

Из граничного условия будем иметь

> subs(r=a, rhs(%))=p(z);

$$\int_0^{\infty} \text{Bessel}(0, va) B(v) \sin(vz) dv = \frac{1}{2} (V_2 - V_1) \text{signum}(z)$$

то есть

$$v(a, z) = \int_0^{\infty} B_v \sin(vz) I_0(va) dv = \frac{V_2 - V_1}{2} \text{signum}(z)$$

Воспользуемся известной формулой

$$\text{signum}(z) = \frac{2}{\pi} \int_0^{\infty} \frac{\sin(vz)}{v} dv$$

Действительно,

> assume(z, real):

$$2/\text{Pi} * \text{Int}(\sin(\text{nu} * z) / \text{nu}, \text{nu}=0..infinity) =$$

$$2/\text{Pi} * \text{int}(\sin(\text{nu} * z) / \text{nu}, \text{nu}=0..infinity);$$

$$\frac{2}{\pi} \int_0^{\infty} \frac{\sin(v z)}{v} dv = \text{signum}(z)$$

Тогда получаем

$$> \text{subs}(r=a, v) = 1/2 * (V[2] - V[1]) * \text{lhs}(\%);$$

$$\int_0^{\infty} B(v) \sin(v z) \text{BesselI}(0, v a) dv = \frac{V_2 - V_1}{\pi} \int_0^{\infty} \frac{\sin(v z)}{v} dv$$

Откуда находим

$$> B(\text{nu}) := (V[2] - V[1]) / \text{Pi} / \text{nu} / \text{BesselI}(0, \text{nu} * a);$$

$$B(v) := \frac{V_2 - V_1}{\pi v \text{BesselI}(0, v a)}$$

Итак, мы получили

$$B_v = \frac{V_2 - V_1}{\pi v I_0(va)}$$

Следовательно,

$$> v := \text{unapply}(v, r, z) : 'v(r, z)' = v(r, z);$$

$$v(r, z) = \int_0^{\infty} \frac{(V_2 - V_1) \sin(v z) \text{BesselI}(0, v r)}{\pi v \text{BesselI}(0, v a)} dv$$

$$> 'u(r, z)' = u(r, z);$$

$$u(r, z) = \frac{1}{2} V_1 + \frac{1}{2} V_2 + \int_0^{\infty} \frac{(V_2 - V_1) \sin(v z) \text{BesselI}(0, v r)}{\pi v \text{BesselI}(0, v a)} dv$$

Таким образом, решение нашей задачи дается формулами

$$v(r, z) = \frac{V_2 - V_1}{\pi} \int_0^{\infty} \frac{I_0(vr) \sin(vz)}{I_0(va) v} dv,$$

$$u(r, z) = \frac{V_2 + V_1}{2} + \frac{V_2 - V_1}{\pi} \int_0^{\infty} \frac{I_0(vr) \sin(vz)}{I_0(va) v} dv.$$

Задача 6.7. Торцы полубесконечного цилиндра ($0 \leq r \leq a$, $0 \leq z < \infty$) поддерживаются при постоянной температуре T_0 , а боковая поверхность находится при температуре, равной нулю. Получить стационарное распределение температуры в цилиндре, разлагая искомую функцию в синус-интеграл Фурье по переменной z .

Решение. Сформулируем математическую задачу: найти функцию $u(r, z)$, удовлетворяющую уравнению Лапласа и дополнительным условиям

$$u|_{z \rightarrow \infty} < \infty, \quad u|_{r \rightarrow 0} < \infty, \quad u|_{r=a} = 0, \quad u|_{z=0} = T_0.$$

Избавимся от неоднородности в граничном условии $u|_{z=0} = T_0$. Для этого будем искать решение задачи в виде $u(r, z) = T_0 + w(r, z)$.

Выполним все вычисления в Maple. Задаем уравнение и дополнительные условия

```
> restart;with(linalg):
Warning, the protected names norm and trace have been
redefined and unprotected
> pde:=laplacian(u(r,z),[r,phi,z],coords=cylindrical)=0;
bc1:=u(a,z)=0;bc2:=u(r,0)=T[0];
```

$$pde := \frac{\left(\frac{\partial}{\partial r} u(r, z)\right) + r \left(\frac{\partial^2}{\partial r^2} u(r, z)\right) + r \left(\frac{\partial^2}{\partial z^2} u(r, z)\right)}{r} = 0$$

$$bc1 := u(a, z) = 0$$

$$bc2 := u(r, 0) = T_0$$

Определяем в Maple функцию $u(r, z) = T_0 + w(r, z)$ и получаем задачу для функции $w(r, z)$

```
> u:=(r,z)->T[0]+w(r,z);
u := (r, z) -> T_0 + w(r, z)
```

```
> pde1:=pde;
```

$$pde1 := \frac{\left(\frac{\partial}{\partial r} w(r, z)\right) + r \left(\frac{\partial^2}{\partial r^2} w(r, z)\right) + r \left(\frac{\partial^2}{\partial z^2} w(r, z)\right)}{r} = 0$$

```
> bc1;bc_1:=isolate(%,w(a,z));
```

$$T_0 + w(a, z) = 0$$

$$bc_1 := w(a, z) = -T_0$$

```
> bc2;bc_2:=isolate(%,w(r,0));
```

$$T_0 + w(r, 0) = T_0$$

$$bc_2 := w(r, 0) = 0$$

Итак, мы имеем следующую задачу:

$$\frac{\partial^2 w}{\partial r^2} + \frac{1}{r} \frac{\partial w}{\partial r} + \frac{\partial^2 w}{\partial z^2} = 0,$$

$$w|_{z \rightarrow \infty} < \infty, \quad w|_{r \rightarrow 0} < \infty, \quad w|_{r=a} = -T_0, \quad w|_{z=0} = 0.$$

Выполняем разделение переменных и формулируем соответствующую задачу Штурма-Лиувилля

```
> res := pdsolve (pde1, HINT=R (r) * Z (z) );
```

```
res := (w (r, z) = R (r) Z (z)) &where
```

$$\left[\left\{ \frac{d^2}{dr^2} R(r) = R(r) _c_1 - \frac{d}{dr} R(r), \frac{d^2}{dz^2} Z(z) = -Z(z) _c_1 \right\} \right]$$

```
> ode1 := op (1, op (1, op (2, res) ) ) ;
```

```
ode2 := op (2, op (1, op (2, res) ) ) ;
```

$$ode1 := \frac{d^2}{dr^2} R(r) = R(r) _c_1 - \frac{d}{dr} R(r)$$

$$ode2 := \frac{d^2}{dz^2} Z(z) = -Z(z) _c_1$$

```
> ode1 := subs (_c [1]=lambda, ode1) ;
```

```
ode2 := subs (_c [1]=lambda, ode2) ;
```

$$ode1 := \frac{d^2}{dr^2} R(r) = R(r) \lambda - \frac{d}{dr} R(r)$$

$$ode2 := \frac{d^2}{dz^2} Z(z) = -Z(z) \lambda$$

```
> ode1 := collect (DEtools [DENormal] (ode1, r, R (r) ), R (r) ) ;
```

```
ode2 := collect (DEtools [DENormal] (ode2, z, Z (z) ), Z (z) ) ;
```

$$ode1 := -r \lambda R(r) + \left(\frac{d}{dr} R(r) \right) + \left(\frac{d^2}{dr^2} R(r) \right) r = 0$$

$$ode2 := \left(\frac{d^2}{dz^2} Z(z) \right) + Z(z) \lambda = 0$$

Итак, задача Штурма-Лиувилля имеет вид

$$Z'' + \lambda Z = 0, \quad Z|_{z \rightarrow \infty} < \infty, \quad Z|_{z=0} = 0.$$

Решаем уравнение Штурма-Лиувилля с учетом граничного условия на торце $z = 0$

```
> dsolve ({ode2, Z (0)=0}, Z (z) ) ;
```

$$Z(z) = {}_0C1 \sin(\sqrt{\lambda} z)$$

С учетом ограниченности функции на бесконечности мы должны считать, что $\lambda > 0$. Таким образом, собственные значения образуют непрерывный спектр: $\lambda = \lambda_\nu = \nu^2$, $0 < \nu < \infty$. Соответствующие им собственные функции будут $Z = Z_\nu(z) = \sin(\nu z)$, или, в Maple

```
> Z := (nu, z) -> sin(nu*z);
```

$$Z := (\nu, z) \rightarrow \sin(\nu z)$$

Решаем второе уравнение

```
> dsolve(subs(lambda=nu^2, odel), R(r));
```

$$R(r) = {}_0C1 \text{BesselI}(0, \nu r) + {}_0C2 \text{BesselK}(0, \nu r)$$

В силу ограниченности функции в нуле мы должны считать ${}_0C2=0$. Следовательно, $R(r) = I_0(\nu r)$, или, в Maple

```
> R := (nu, r) -> BesselI(0, nu*r);
```

$$R := (\nu, r) \rightarrow \text{BesselI}(0, \nu r)$$

Таким образом, решение задачи дается интегралом

$$w(r, t) = \int_0^\infty C_\nu I_0(\nu r) \sin(\nu z) d\nu$$

или, в Maple

```
> v := Int(C(nu)*Z(nu, z)*R(nu, r), nu=0..infinity);
```

$$v := \int_0^\infty C(\nu) \sin(\nu z) \text{BesselI}(0, \nu r) d\nu$$

Подставляем в граничное условие, получим

```
> subs(r=a, v) = -T[0];
```

$$\int_0^\infty C(\nu) \sin(\nu z) \text{BesselI}(0, \nu a) d\nu = -T_0$$

Воспользуемся формулой

$$1 = \frac{2}{\pi} \int_0^\infty \frac{\sin(\nu z)}{\nu} d\nu, \quad z > 0$$

Действительно,

```
> assume(z>0):
```

$$2/\text{Pi} * \text{Int}(\sin(\text{nu}*z)/\text{nu}, \text{nu}=0..infinity) =$$

$$2/\text{Pi} * \text{int}(\sin(\text{nu}*z)/\text{nu}, \text{nu}=0..infinity);$$

$$\frac{2}{\pi} \int_0^{\infty} \frac{\sin(v z)}{v} dv = 1$$

Тогда получаем

> subs (r=a, v)=-T[0]*lhs(%);

$$\int_0^{\infty} C(v) \sin(v z) \text{BesselI}(0, v a) dv = - \left(\frac{2 T_0}{\pi} \int_0^{\infty} \frac{\sin(v z)}{v} dv \right)$$

Откуда находим

> C(nu) :=-2*T[0]/Pi/nu/BesselI(0, nu*a);

$$C(v) := - \frac{2 T_0}{\pi v \text{BesselI}(0, v a)}$$

Итак, мы нашли коэффициенты разложения

$$C_v = - \frac{2T_0}{\pi I_0(va)}.$$

Следовательно,

> w:=unapply(v, r, z) : 'w(r, z)'=w(r, z);

$$w(r, z) = \int_0^{\infty} - \frac{2 T_0 \sin(v z) \text{BesselI}(0, v r)}{\pi v \text{BesselI}(0, v a)} dv$$

> 'u(r, z)'=u(r, z);

$$u(r, z) = T_0 + \int_0^{\infty} - \frac{2 T_0 \sin(v z) \text{BesselI}(0, v r)}{\pi v \text{BesselI}(0, v a)} dv$$

Таким образом, окончательно, решение нашей задачи дается формулами

$$v(r, z) = - \frac{2T_0}{\pi} \int_0^{\infty} \frac{I_0(vr)}{I_0(va)} \frac{\sin(vz)}{v} dv,$$

$$u(r, z) = T_0 \left[1 - \frac{2}{\pi} \int_0^{\infty} \frac{I_0(vr)}{I_0(va)} \frac{\sin(vz)}{v} dv \right].$$

Отметим, что эту задачу можно решать и с помощью синус-преобразования Фурье. Для этого умножаем исходное уравнение на $\sin(vz)$ и интегрируем по области $[0, +\infty)$. При этом будем иметь

$$\int_0^{\infty} \frac{1}{r} \frac{\partial}{\partial r} \left(r \frac{\partial u}{\partial r} \right) \sin(vz) dz = \frac{1}{r} \frac{\partial}{\partial r} \left(r \frac{\partial}{\partial r} \int_0^{\infty} u \sin(vz) dz \right) = \frac{1}{r} (r\tilde{u}')',$$

$$\int_0^{\infty} \frac{\partial^2 u}{\partial z^2} \sin(vz) dz = \left[\frac{\partial u}{\partial z} \sin(vz) - vu \cos(vz) \right]_{z=0}^{z=\infty} - v^2 \tilde{u},$$

где $\tilde{u}(r) = \int_0^{\infty} u(r, z) \sin(vz) dz$.

Потребуем дополнительно, чтобы $\frac{\partial u}{\partial z} \Big|_{z=\infty} \rightarrow 0$ и $u \Big|_{z=\infty} \rightarrow 0$. Это не противоречит тепловой задаче. Тогда получаем уравнение для синус-трансформанты Фурье

$$\frac{1}{r} (r\tilde{u}')' - v^2 \tilde{u} + vT_0 = 0.$$

Откуда находим

$$\tilde{u}(r) = AI_0(vr) + BK_0(vr) + \frac{T_0}{v}.$$

В силу ограниченности функции в нуле получаем $B = 0$, а из условия при $r = a$ находим

$$AI_0(va) + \frac{T_0}{v} = 0 \Rightarrow A = -\frac{T_0}{vI_0(va)}.$$

Таким образом,

$$\tilde{u}(r) = -\frac{T_0}{vI_0(va)} I_0(vr) + \frac{T_0}{v}.$$

Откуда

$$u(r, z) = -\frac{2}{\pi} \int_0^{\infty} \frac{T_0}{vI_0(va)} I_0(vr) \sin(vz) dv + \frac{2T_0}{\pi} \int_0^{\infty} \frac{\sin(vz)}{v} dv.$$

Или, учитывая формулу

$$1 = \frac{2}{\pi} \int_0^{\infty} \frac{\sin(vz)}{v} dv, \quad z > 0,$$

находим

$$u(r, z) = T_0 \left[1 - \frac{2}{\pi} \int_0^{\infty} \frac{I_0(vr)}{I_0(va)} \frac{\sin(vz)}{v} dv \right].$$

Задача 6.8. Найти распределение электростатического поля на оси плоской электронной линзы, конструкция которой изображена на рис. 6.3.




Рис. 6.3. Плоская электронная линза

Решение. Сформулируем математическую задачу: найти функцию $u(x, y)$, удовлетворяющую уравнению Лапласа и дополнительным условиям

$$u|_{y=h} = u|_{y=-h} = p(x), \quad u|_{x \rightarrow \pm\infty} < \infty, \quad p(x) = \begin{cases} V_1, & |x| < a, \\ V_2, & |x| > a. \end{cases}$$

Решение задачи будем искать в виде

$$u(x, y) = V_2 + w(x, y). \quad (*)$$

Выполним все вычисления в Maple. Задаем уравнение и дополнительные условия

```
> restart;
> with(linalg):
Warning, the protected names norm and trace have been
redefined and unprotected
> pde:=laplacian(u(x,y), [x,y])=0;
bc:=[u(x,h)=p(x), u(x,-h)=p(x)];
'p(x)'=piecewise(abs(x)<a,V[1],abs(x)>a,V[2]);
```

$$pde := \left(\frac{\partial^2}{\partial x^2} u(x, y) \right) + \left(\frac{\partial^2}{\partial y^2} u(x, y) \right) = 0$$

$$bc := [u(x, h) = p(x), u(x, -h) = p(x)]$$

$$p(x) = \begin{cases} V_1 & |x| < a \\ V_2 & a < |x| \end{cases}$$

Ищем решение задачи в виде

```
> u := (x, y) -> V[2] + w(x, y);
u := (x, y) -> V_2 + w(x, y)
```

Тогда для функции $w(x, y)$ будем иметь следующую задачу

> pde1:=pde;

$$pde1 := \left(\frac{\partial^2}{\partial x^2} w(x, y) \right) + \left(\frac{\partial^2}{\partial y^2} w(x, y) \right) = 0$$

> bc1:=[w(x, h)=p(x), w(x, -h)=p(x)];

p:=x->piecewise(abs(x)<a, V[1]-V[2], abs(x)>a, 0):

'p(x)'=p(x);

$$bc1 := [w(x, h) = p(x), w(x, -h) = p(x)]$$

$$p(x) = \begin{cases} V_1 - V_2 & |x| < a \\ 0 & a < |x| \end{cases}$$

Выполняем разделение переменных:

> res:=pdsolve(pde1, HINT=X(x)*Y(y));

res := (w(x, y) = X(x) Y(y)) &where

$$\left[\left\{ \frac{d^2}{dy^2} Y(y) = -c_1 Y(y), \frac{d^2}{dx^2} X(x) = -c_1 X(x) \right\} \right]$$

> ode1:=op(1, op(1, (op(2, res))));

ode2:=op(2, op(1, (op(2, res))));

$$ode1 := \frac{d^2}{dx^2} X(x) = -c_1 X(x)$$

$$ode2 := \frac{d^2}{dy^2} Y(y) = -c_1 Y(y)$$

> ode1:=subs(_c[1]=-lambda, ode1);

ode2:=subs(_c[1]=-lambda, ode2);

$$ode1 := \frac{d^2}{dx^2} X(x) = -\lambda X(x)$$

$$ode2 := \frac{d^2}{dy^2} Y(y) = \lambda Y(y)$$

> ode1:=collect(DEtools[DEnormal](ode1, x, X(x)), X(x));

ode2:=collect(DEtools[DEnormal](ode2, y, Y(y)), Y(y));

$$ode1 := \left(\frac{d^2}{dx^2} X(x) \right) + \lambda X(x) = 0$$

$$ode2 := \left(\frac{d^2}{dy^2} Y(y) \right) - \lambda Y(y) = 0$$

Таким образом, мы приходим к следующей задаче Штурма-Лиувилля

$$X'' + \lambda X = 0, \quad X|_{x \rightarrow \pm\infty} = O(1).$$

Решаем дифференциальное уравнение

> dsolve(ode1, X(x));

$$X(x) = _C1 \sin(\sqrt{\lambda} x) + _C2 \cos(\sqrt{\lambda} x)$$

Из условия ограниченности на бесконечности следует, что $\lambda > 0$, а тогда собственные значения будут $\lambda = \lambda_v = v^2$, $0 < v < \infty$; соответствующие им собственные функции

> X := (nu, x) -> A(nu) * cos(nu*x) + B(nu) * sin(nu*x);

$$X := (v, x) \rightarrow A(v) \cos(v x) + B(v) \sin(v x)$$

Решаем второе уравнение

> dsolve(subs(lambda=nu^2, ode1), Y(y));

$$Y(y) = _C1 e^{(-v y)} + _C2 e^{(v y)}$$

В силу симметрии решения по переменной y выбираем это решение в виде

> Y := (nu, y) -> cosh(nu*y);

$$Y := (v, y) \rightarrow \cosh(v y)$$

Таким образом, решение задачи дается интегралом

> v := Int(X(nu, x) * Y(nu, y), nu=0..infinity);

$$v := \int_0^{\infty} (A(v\sim) \cos(v\sim x) + B(v\sim) \sin(v\sim x)) \cosh(v\sim y) dv\sim$$

Подставляем в граничное условие, получим

> subs(y=h, v) = 'p(x)';

$$\int_0^{\infty} (A(v\sim) \cos(v\sim x) + B(v\sim) \sin(v\sim x)) \cosh(v\sim h) dv\sim = p(x)$$

Откуда находим

> assume(a>0):

$$eq1 := A(nu) * \cosh(nu*h) =$$

$$1/Pi * \int (p(x) * \cos(nu*x), x=-infinity..infinity);$$

$$eq2 := B(nu) * \cosh(nu*h) =$$

$$1/Pi * \int (p(x) * \sin(nu*x), x=-infinity..infinity);$$

$$eq1 := A(v\sim) \cosh(v\sim h) = \frac{2(V_1 - V_2) \sin(v\sim a\sim)}{\pi v\sim}$$

$$eq2 := B(v\sim) \cosh(v\sim h) = 0$$

Следовательно,

> isolate(eq1, A(nu)) : assign(%) : 'A(nu)' = A(nu) ; B(nu) := 0 ;

$$A(v) = \frac{2(V_1 - V_2) \sin(v\sim a\sim)}{\pi v\sim \cosh(v\sim h)}$$

$$B(v\sim) := 0$$

А тогда

> expand(v) ;

$$\frac{2(V_1 - V_2)}{\pi} \int_0^{\infty} \frac{\sin(v a\sim) \cos(v x) \cosh(v y)}{v \cosh(v h)} dv$$

> w := unapply(%, x, y) : 'w(x, y)' = w(x, y) ;

$$w(x, y) = \frac{2(V_1 - V_2)}{\pi} \int_0^{\infty} \frac{\sin(v a\sim) \cos(v x) \cosh(v y)}{v \cosh(v h)} dv$$

> 'u(x, y)' = u(x, y) ;

$$u(x, y) = V_2 + \frac{2(V_1 - V_2)}{\pi} \int_0^{\infty} \frac{\sin(v a\sim) \cos(v x) \cosh(v y)}{v \cosh(v h)} dv$$

Итак, решение задачи дается формулой

$$w(x, y) = \frac{2(V_1 - V_2)}{\pi} \int_0^{\infty} \frac{\operatorname{ch}(vy) \sin(va)}{\operatorname{ch}(vh) v} \cos(vx) dv,$$

$$u(x, y) = V_2 + \frac{2(V_1 - V_2)}{\pi} \int_0^{\infty} \frac{\operatorname{ch}(vy) \sin(va)}{\operatorname{ch}(vh) v} \cos(vx) dv.$$

Задача 6.9. Тепловой поток Q поступает через участок $|x| < a$ плоской границы полубесконечного тела (рис. 6.4). Найти стационарное распределение температуры в теле, считая поток равномерно распределенным и предполагая, что поверхность тела излучает тепло в окружающую среду по закону Ньютона.




Рис. 6.4. К задаче 6.9

Решение. Сформулируем математическую задачу: найти функцию $u(x, y)$, удовлетворяющую уравнению Лапласа и дополнительным условиям

$$\frac{\partial u}{\partial y} - hu \Big|_{y=0} = p(x), \quad u|_{x \rightarrow \pm\infty} = O(1), \quad p(x) = \begin{cases} -\frac{q}{k}, & |x| \leq a, \\ 0, & |x| > a, \end{cases}$$

где $q = \frac{Q}{2a}$.

Выполним все вычисления в Maple. Задаем уравнение и дополнительные условия

```
> restart;
> pde:=linalg[laplacian](u(x,y),[x,y])=0;
      pde := (∂²/∂x² u(x,y)) + (∂²/∂y² u(x,y)) = 0
> bc:=D(u)(x,0)-h*u(x,0)='p(x)';
      'p(x)'=piecewise(abs(x)<=a,-q/k,abs(x)>a,0);
      bc := D(u)(x,0) - h u(x,0) = p(x)
      p(x) = { -q/k   |x| ≤ a
               0     a < |x|
```

Выполняем разделение переменных

```
> res:=pdsolve(pde,HINT=X(x)*Y(y));
      res := (u(x,y) = X(x) Y(y)) &where
      [ { d²/dy² Y(y) = -c₁ Y(y), d²/dx² X(x) = -c₁ X(x) } ]
> ode1:=op(1,op(1,op(2,res)));
```

```
ode2:=op(2,op(1,(op(2,res))));
```

$$ode1 := \frac{d^2}{dy^2} Y(y) = -c_1 Y(y)$$

$$ode2 := \frac{d^2}{dx^2} X(x) = -c_1 X(x)$$

```
> ode1:=subs(_c[1]=-lambda,ode1);
```

```
ode2:=subs(_c[1]=-lambda,ode2);
```

$$ode1 := \frac{d^2}{dy^2} Y(y) = \lambda Y(y)$$

$$ode2 := \frac{d^2}{dx^2} X(x) = -\lambda X(x)$$

```
> ode2:=collect(DEtools[DEnormal](ode2,x,X(x)),X(x));
```

```
ode1:=collect(DEtools[DEnormal](ode1,y,Y(y)),Y(y));
```

$$ode2 := \left(\frac{d^2}{dx^2} X(x) \right) + \lambda X(x) = 0$$

$$ode1 := \left(\frac{d^2}{dy^2} Y(y) \right) - \lambda Y(y) = 0$$

Таким образом, имеем следующую задачу Штурма-Лиувилля

$$X'' + \lambda X = 0, \quad X|_{x \rightarrow \pm\infty} = O(1).$$

Решаем дифференциальное уравнение

```
> dsolve(ode2,X(x));
```

$$X(x) = _C1 \sin(\sqrt{\lambda} x) + _C2 \cos(\sqrt{\lambda} x)$$

Из условия ограниченности на бесконечности следует, что $\lambda > 0$; собственные значения будут $\lambda = \lambda_\nu = \nu^2$, $0 < \nu < \infty$, а соответствующие собственные функции

```
> X:=(nu,x)->A(nu)*cos(nu*x)+B(nu)*sin(nu*x);
```

$$X := (\nu, x) \rightarrow A(\nu) \cos(\nu x) + B(\nu) \sin(\nu x)$$

Решаем второе уравнение

```
> dsolve(subs(lambda=nu^2,ode1),Y(y));
```

$$Y(y) = _C1 e^{(-\nu y)} + _C2 e^{(\nu y)}$$

В силу ограниченности функции на бесконечности выбираем решение в виде ($\nu > 0, y > 0$):

```
> Y:=(nu,y)->exp(-nu*y);
```


$$Y := (v, y) \rightarrow e^{(-v y)}$$

Таким образом, решение задачи дается интегралом

> v:=Int(X(nu, x)*Y(nu, y), nu=0..infinity);

$$v := \int_0^{\infty} (A(v) \cos(v x) + B(v) \sin(v x)) e^{(-v y)} dv$$

Подставляем $y = 0$ в это выражение

> combine(value(subs(y=0, diff(v, y)-h*v)));

$$\int_0^{\infty} -v A(v) \cos(v x) - v B(v) \sin(v x) - h A(v) \cos(v x) - h B(v) \sin(v x) dv$$

Приведем подобные. Для этого выделяем подынтегральную функцию и упростим

> f:=op(1, %);

f:=-v A(v) cos(v x) - v B(v) sin(v x) - h A(v) cos(v x) - h B(v) sin(v x)

> f:=collect(f, cos(nu*x)); f:=collect(f, sin(nu*x));

f:=(-v A(v) - h A(v)) cos(v x) - v B(v) sin(v x) - h B(v) sin(v x)

f:=(-v B(v) - h B(v)) sin(v x) + (-v A(v) - h A(v)) cos(v x)

Таким образом, получаем

> Int(f, nu=0..infinity)='p(x)';

$$\int_0^{\infty} (-v B(v) - h B(v)) \sin(v x) + (-v A(v) - h A(v)) \cos(v x) dv = p(x)$$

Откуда будем иметь систему

> eq1:=op(1, op(1, f))=1/Pi*int((-q/k)*sin(nu*x), x=-a..a);

eq2:=op(1, op(2, f))=1/Pi*int((-q/k)*cos(nu*x), x=-a..a);

$$eq1 := -v B(v) - h B(v) = 0$$

$$eq2 := -v A(v) - h A(v) = -\frac{2 q \sin(v a)}{\pi k v}$$

Решаем эту систему

> solve({eq1, eq2}, {A(nu), B(nu)});

$$\{A(v) = \frac{2 q \sin(v a)}{\pi k v (v + h)}, B(v) = 0\}$$

> assign(%):

В результате получаем

> 'u(x, y)'=expand(subs(q=Q/2/a, v));

$$u(x, y) = \frac{2q}{\pi k} \int_0^{\infty} \frac{\sin(v a) \cos(v x)}{v(v+h) e^{(v y)}} dv$$

Итак, решение задачи дается формулой $\left(q = \frac{Q}{2a} \right)$

$$u(x, y) = \frac{Q}{\pi a k} \int_0^{\infty} \frac{\sin(va)}{v(v+h)} e^{-vy} \cos(vx) dv.$$

Задача 6.10. Плоская граница полубесконечного тела, начиная с момента времени $t = 0$, поддерживается при температуре, изменяющейся по закону $u|_{x=0} = f(\tau)$, где $\tau = \frac{kt}{c\rho}$ (k – коэффициент теплопроводности, c – удельная теплоемкость, ρ – плотность вещества). Найти значение температуры в каждой точке тела, принимая его начальную температуру равной нулю.

Решение. Сформулируем математическую задачу: найти функцию $u(x, \tau)$, удовлетворяющую уравнению теплопроводности

$$\frac{\partial^2 u}{\partial x^2} - \frac{\partial u}{\partial \tau} = 0, \quad x > 0, \quad \tau > 0$$

и дополнительным условиям

$$u|_{x \rightarrow \infty} = O(1), \quad u|_{x=0} = f(\tau), \quad u|_{\tau=0} = 0.$$

Решаем задачу в системе Maple с помощью преобразования Лапласа. Определяем задачу:

```
> restart;
> pde:=diff(u(x,tau),x$2)-diff(u(x,tau),tau)=0; x>0,tau>0;
bc:=u(0,tau)=f(tau);
ic:=u(x,0)=0;
```

$$pde := \left(\frac{\partial^2}{\partial x^2} u(x, \tau) \right) - \left(\frac{\partial}{\partial \tau} u(x, \tau) \right) = 0$$

$$bc := u(0, \tau) = f(\tau)$$

$$ic := u(x, 0) = 0$$

Подключаем необходимые команды и применяем преобразование Лапласа к заданному уравнению теплопроводности

```
> with(inttrans,laplace,invlaplace);
[laplace,invlaplace]
```

> laplace (pde, tau, p) ;

$$\left(\frac{\partial^2}{\partial x^2} \text{laplace} (u(x, \tau), \tau, p) \right) - p \text{laplace} (u(x, \tau), \tau, p) + u(x, 0) = 0$$

> ode:=subs (laplace (u (x, tau) , tau, p)=U (x) , ic, %) ;

$$ode := \left(\frac{d^2}{dx^2} U(x) \right) - p U(x) = 0$$

Решаем уравнение для трансформанты Лапласа

> res:=dsolve (ode, U (x)) ; assign (res) ; U (x) ;

$$res := U(x) = _C1 e^{(\sqrt{p} x)} + _C2 e^{(-\sqrt{p} x)} \\ _C1 e^{(\sqrt{p} x)} + _C2 e^{(-\sqrt{p} x)}$$

С учетом условия ограниченности на бесконечности получаем

> U:=subs (_C1=0, U (x)) ;

$$U := _C2 e^{(-\sqrt{p} x)}$$

Преобразовываем граничное условие

> laplace (bc, tau, p) ;

$$\text{laplace} (u(0, \tau), \tau, p) = \text{laplace} (f(\tau), \tau, p)$$

> value (subs (laplace (u (0, tau) , tau, p) =

subs (x=0, U) , lhs (%)) =rhs (%) ;

$$_C2 = \text{laplace} (f(\tau), \tau, p)$$

Определяем константу

> solve (% , { _C2 }) ; assign (%) ; U ;

$$\{ _C2 = \text{laplace} (f(\tau), \tau, p) \}$$

$$\text{laplace} (f(\tau), \tau, p) e^{(-\sqrt{p} x)}$$

и окончательно находим трансформанту Лапласа

> U:=subs (U (x)=laplace (u (x, tau) , tau, p) , %) ;

$$U := \text{laplace} (f(\tau), \tau, p) e^{(-\sqrt{p} x)}$$

Вычисляем обратное преобразование Лапласа

> invlaplace (U, p, tau) ;

$$\int_0^{\tau} \frac{1}{2} \frac{f(_UI) x \sim e^{\left(-\frac{x^2}{4(\tau - _UI)} \right)}}{\sqrt{\pi} (\tau - _UI)^{(3/2)}} d_UI$$

> expand(%);

$$\frac{1}{2} \left(\frac{x}{\sqrt{\pi}} \int_0^{\tau} \frac{f(U) e^{-\frac{x^2}{4(\tau-U)}}}{(\tau-U)^{3/2}} dU \right)$$

Таким образом, решение задачи получено в форме

$$u(x, \tau) = \frac{x}{2\sqrt{\pi}} \int_0^{\tau} \frac{f(\xi) e^{-\frac{x^2}{4(\tau-\xi)}}}{(\tau-\xi)^{3/2}} d\xi.$$

Этот интеграл можно преобразовать с помощью замены переменной по формуле

$$y^2 = \frac{x^2}{4(\tau-\xi)}.$$

Выполним эту замену в Maple

> convert(% , Int):

assume(x>0): assume(y>0):

simplify(student[changevar](x^2/4/(tau-U)=y^2, %, y));

$$\frac{1}{2} \left(\frac{x}{\sqrt{\pi}} \int_{\frac{x}{2}\sqrt{\frac{1}{\tau}}}^{\infty} \frac{4 f\left(\frac{-x^2 + 4 y^2 \tau}{4 y^2}\right) e^{-y^2}}{x} dy \right)$$

> expand(%);

$$\frac{2}{\sqrt{\pi}} \int_{\frac{x}{2}\sqrt{\frac{1}{\tau}}}^{\infty} \frac{f\left(\frac{-x^2 + 4 y^2 \tau}{4 y^2}\right)}{e^{y^2}} dy$$

Итак, решение задачи можно представить и в таком виде

$$u(x, \tau) = \frac{2}{\sqrt{\pi}} \int_{\frac{x}{2\sqrt{\tau}}}^{\infty} f\left(\tau - \frac{x^2}{4y^2}\right) e^{-y^2} dy.$$

Рассмотрим некоторые частные случаи. Пусть, например, начальная функция — константа, то есть $f(\tau) = T_0$. Тогда определим наше решение как функцию в Maple

> u:=unapply(%, (x, tau));

$$u := (x, \tau) \rightarrow \frac{2}{\sqrt{\pi}} \int_{1/2 x \sqrt{1/\tau}}^{\infty} \frac{f\left(\frac{1-x^2+4y^2\tau}{4y^2}\right)}{e^{(y^2)}} dy$$

и найдем распределение температуры в этом частном случае

> f:=tau->T[0];

$$f := \tau \rightarrow T_0$$

> simplify(value(u(x, tau)));

$$-T_0 \left(-1 + \operatorname{erf}\left(\frac{x \sqrt{1/\tau}}{2}\right) \right)$$

То есть, получаем

$$u(x, \tau) = T_0 \left[1 - \Phi\left(\frac{x}{2\sqrt{\tau}}\right) \right],$$

где $\Phi(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-z^2} dz$ — интеграл вероятности.

Пусть, например, начальная температура изменяется линейно, то есть $f(\tau) = A\tau$.

Тогда

> f:=tau->A*tau;

$$f := \tau \rightarrow A \tau$$

> expand(value(u(x, tau)));

$$\frac{A x^2}{2} + A \tau - \frac{A x e^{-\left(\frac{x^2}{4\tau}\right)}}{\sqrt{\pi} \sqrt{\frac{1}{\tau}}} - A \tau \operatorname{erf}\left(\frac{x \sqrt{1/\tau}}{2}\right) - \frac{1}{2} A x^2 \operatorname{erf}\left(\frac{x \sqrt{1/\tau}}{2}\right)$$

> collect(%, erf);

$$\left(-A \tau - \frac{1}{2} A x^2\right) \operatorname{erf}\left(\frac{x \sqrt{1/\tau}}{2}\right) + \frac{A x^2}{2} + A \tau - \frac{A x e^{-\left(\frac{x^2}{4\tau}\right)}}{\sqrt{\pi} \sqrt{\frac{1}{\tau}}}$$

То есть в этом случае имеем

$$u(x, \tau) = -A \left(\tau + \frac{x^2}{2} \right) \Phi \left(\frac{x}{2\sqrt{\tau}} \right) + A \left(\tau + \frac{x^2}{2} \right) - \frac{Ax\sqrt{\tau}}{\sqrt{\pi}} e^{-\frac{x^2}{4\tau}}.$$

Упражнения

1. Решить задачу

$$\frac{\partial u}{\partial t} - a^2 \frac{\partial^2 u}{\partial x^2} = 0, \quad u|_{t=0} = \varphi(x), \quad -\infty < x < \infty, \quad t > 0.$$

Ответ. $u(x, t) = \frac{1}{2a\sqrt{\pi t}} \int_{-\infty}^{+\infty} \varphi(\xi) e^{-\frac{(\xi-x)^2}{4a^2 t}} d\xi.$

2. Решить задачу Дирихле для уравнения Лапласа для верхней полуплоскости.

Ответ. $u(x, y) = \frac{y}{\pi} \int_{-\infty}^{+\infty} \frac{\varphi(\xi)}{(x-\xi)^2 + y^2} d\xi.$

3. Решить задачу

$$\frac{\partial^2 u}{\partial t^2} + \Delta^2 u = xy e^t, \quad u|_{t=0} = 0, \quad \frac{\partial u}{\partial t}|_{t=0} = 0, \quad -\infty < x, y, z < \infty, \quad t > 0.$$

Ответ. $u(x, y, z, t) = 3x^2 y^2 z^2 t^2 - 2(x^2 + y^2 + z^2)t^4.$

4. Решить задачу

$$\frac{\partial^2 u}{\partial t^2} + \frac{\partial^4 u}{\partial x^4} = 0, \quad u|_{t=0} = \cos(x^2), \quad \frac{\partial u}{\partial t}|_{t=0} = 0, \quad -\infty < x < \infty, \quad 0 < t < \frac{1}{4}.$$

Ответ. $u(x, t) = \frac{1}{2\sqrt{1+4t}} \cos\left(\frac{x^2}{1+4t}\right) + \frac{1}{2\sqrt{1-4t}} \cos\left(\frac{x^2}{1-4t}\right).$

5. Решить следующую задачу (задача о радиальных колебаниях газа):

$$\frac{1}{r^2} \frac{\partial}{\partial r} \left(r^2 \frac{\partial u}{\partial r} \right) - \frac{1}{v^2} \frac{\partial^2 u}{\partial t^2} = 0, \quad 0 < r < \infty, \quad t > 0;$$

$$u|_{t=0} = \varphi(r), \quad \frac{\partial u}{\partial t}|_{t=0} = 0;$$

$$u|_{r \rightarrow 0} - \text{ограничена}, \quad u|_{r \rightarrow \infty} - \text{ограничена}.$$

Физическая интерпретация задачи: задано начальное распределение плотности газа в пространстве. Требуется проследить дальнейшее распределение плотности — процесс выравнивания давления.

Ответ. $u = \frac{1}{2r} \{ (r + vt) \varphi(r + vt) + (r - vt) \varphi(|r - vt|) \}$.

6. Решить следующую задачу Дирихле для полупространства:

$$\frac{1}{r} \frac{\partial}{\partial r} \left(r \frac{\partial u}{\partial r} \right) + \frac{\partial^2 u}{\partial z^2} = 0, \quad 0 < r < \infty, \quad z > 0,$$

$$u|_{r \rightarrow 0} - \text{ограничена}, \quad u|_{\sqrt{r^2 + z^2} \rightarrow \infty} - \text{ограничена},$$

$$u|_{z=0} = f(r) = \begin{cases} T_0, & r < a, \\ 0, & r > a. \end{cases}$$

Найти значения искомой функции вдоль оси z .

Ответ. $u(r, z) = T_0 a \int_0^{\infty} e^{-vz} J_1(va) J_0(vr) dv$; $u|_{r=0} = T_0 \left(1 - \frac{z}{\sqrt{a^2 + z^2}} \right)$.

7. Решить следующую задачу Дирихле для полуплоскости $y \geq 0$, $-\infty < x < +\infty$:

$$\Delta u \equiv \frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} = 0, \quad y > 0,$$

$$u|_{y=0} = f(x), \quad u|_{\sqrt{x^2 + y^2} \rightarrow \infty} - \text{ограничена}.$$

Рассмотреть случай $f(x) = \begin{cases} u_0, & |x| < a, \\ 0, & |x| > a. \end{cases}$

Ответ. $u(x, y) = \frac{1}{\pi} \int_{+\infty}^{-\infty} f(\xi) \frac{y}{(\xi - x)^2 + y^2} d\xi$;

$$u = \frac{u_0}{\pi} \left[\operatorname{arctg} \left(\frac{a - x}{y} \right) + \operatorname{arctg} \left(\frac{a + x}{y} \right) \right].$$

8. Решить следующую задачу для уравнения теплопроводности (задача об охлаждении полубесконечного тела):

$$\frac{\partial^2 T}{\partial x^2} - \frac{\partial T}{\partial \tau} = 0, \quad 0 < x < +\infty, \quad \tau > 0, \quad \tau = \frac{kt}{c\rho}.$$

$$T|_{x=0} = 0, \quad T|_{x \rightarrow \infty} - \text{ограничена},$$

$$T|_{\tau=0} = \varphi(x).$$

Рассмотреть случай $\varphi(x) = T_0 = \text{const}$.

Ответ. $T(x, \tau) = \frac{1}{2\sqrt{\pi\tau}} \int_0^{\infty} \varphi(\xi) \left[e^{-\frac{(\xi-x)^2}{4\tau}} - e^{-\frac{(\xi+x)^2}{4\tau}} \right] d\xi;$

$T(x, \tau) = T_0 \Phi\left(\frac{x}{2\sqrt{\tau}}\right)$, где $\Phi(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-s^2} ds$ — интеграл вероятности.

9. Решить краевую задачу

$$\frac{\partial u}{\partial t} = a^2 \frac{\partial^2 u}{\partial x^2}, \quad 0 < x, t < +\infty, \quad u|_{x=0} = U_0, \quad u|_{t=0} = 0.$$

Ответ. $u(x, t) = U_0 \left[1 - \Phi\left(\frac{x}{2a\sqrt{t}}\right) \right]$, где $\Phi(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-s^2} ds$ — интеграл вероятности.

10. Решить краевую задачу

$$\frac{\partial u}{\partial t} = a^2 \frac{\partial^2 u}{\partial x^2}, \quad 0 < x, t < +\infty, \quad \frac{\partial u}{\partial x}\Big|_{x=0} = 0, \quad u|_{t=0} = \psi(x) = \begin{cases} U_0, & 0 < x < 1, \\ 0, & 1 < x < +\infty. \end{cases}$$

Ответ. $u(x, t) = \frac{U_0}{2} \left[\Phi\left(\frac{x+1}{2a\sqrt{t}}\right) - \Phi\left(\frac{x-1}{2a\sqrt{t}}\right) \right]$, где $\Phi(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-s^2} ds$ — интеграл вероятности.

11. Тело ограничено изнутри сферической поверхностью радиуса a , снаружи не ограничено. Начальная температура тела равна нулю, а на сферической поверхности температура поддерживается постоянной, равной T_0 . Найти температуру тела в произвольный момент времени t .

Ответ. $u(x, t) = \frac{T_0 a}{r} \left[1 - \Phi\left(\frac{r-a}{2\sqrt{\tau}}\right) \right]$, где $\tau = \frac{kt}{c\rho}$ — приведенное время;

$\Phi(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-s^2} ds$ — интеграл вероятности.

12. Решить краевую задачу

$$\frac{\partial u}{\partial t} = a^2 \frac{\partial^2 u}{\partial x^2}, \quad 0 < x, t < +\infty, \quad \frac{\partial u}{\partial x}\Big|_{x=0} = 0, \quad u|_{t=0} = f(x).$$

Ответ. $u(x, t) = \frac{1}{2a\sqrt{\pi t}} \int_{-\infty}^{+\infty} f(\xi) \left[e^{-\frac{(x-\xi)^2}{4a^2 t}} + e^{-\frac{(x+\xi)^2}{4a^2 t}} \right] d\xi.$

13. Решить краевую задачу

$$\frac{\partial u}{\partial t} = a^2 \frac{\partial^2 u}{\partial x^2}, \quad 0 < x, t < +\infty, \quad \frac{\partial u}{\partial x} - hu \Big|_{x=0} = 0, \quad u|_{t=0} = U_0, \quad h > 0.$$

Ответ. $u(x, t) = U_0 \Phi\left(\frac{x}{2a\sqrt{t}}\right) + e^{hx+h^2 a^2 t} U_0 \left[1 - \Phi\left(\frac{x}{2a\sqrt{t}} + ah\sqrt{t}\right)\right],$

где $\Phi(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-s^2} ds$ — интеграл вероятности.

14. Решить задачу

$$\frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2} + e^{-t} \cos(x), \quad u|_{t=0} = \cos(x).$$

Ответ. $u(x, t) = (1+t)e^{-t} \cos(x).$

15. Решить задачу

$$\frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2} + \sin(t), \quad u|_{t=0} = e^{-x^2}.$$

Ответ. $u(x, t) = 1 - \cos(t) + (1+4t)\frac{1}{2}e^{-\frac{x^2}{1+4t}}.$

16. Два полубесконечных тела из различных материалов, одно из которых нагрето до температуры T_0 , а другое имеет температуру, равную нулю, приводятся в контакт своими плоскими границами начиная с момента времени $t = 0$. Найти закон выравнивания температур.

Ответ. $u_1(x, t) = T_0 - \frac{T_0 k_2 \sqrt{b_2} \left[1 - \Phi\left(\frac{x\sqrt{b_1}}{2\sqrt{t}}\right)\right]}{k_1 \sqrt{b_1} + k_2 \sqrt{b_2}}, \quad x \geq 0, \quad t \geq 0,$

$$u_2(x, t) = \frac{T_0 k_1 \sqrt{b_1} \left[1 + \Phi\left(\frac{x\sqrt{b_2}}{2\sqrt{t}}\right)\right]}{k_1 \sqrt{b_1} + k_2 \sqrt{b_2}}, \quad x \leq 0, \quad t \geq 0,$$

где $\Phi(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-s^2} ds$ — интеграл вероятности.

Указание. Задача сводится к решению следующих уравнений

$$b_1 \frac{\partial u_1}{\partial t} = \frac{\partial^2 u_1}{\partial x^2}, \quad b_1 = \frac{c_1 \rho_1}{k_1}, \quad x > 0, \quad t > 0,$$

$$b_2 \frac{\partial u_2}{\partial t} = \frac{\partial^2 u_2}{\partial x^2}, \quad b_2 = \frac{c_2 \rho_2}{k_2}, \quad x < 0, \quad t > 0.$$

при начальных

$$u_1|_{t=0} = T_0, \quad u_2|_{t=0} = 0.$$

и граничных условиях

$$u_1|_{x=0} = u_2|_{x=0}, \quad k_1 \frac{\partial u_1}{\partial x} \Big|_{x=0} = k_2 \frac{\partial u_2}{\partial x} \Big|_{x=0}.$$

17. Решить задачу Дирихле для уравнения Лапласа в области, представляющей собой полуполосу $D = \{0 < x < \infty, 0 < y < b\}$:

$$\Delta u = 0, \quad (x, y) \in D,$$

$$u|_{x=0} = u_0 = \text{const}, \quad u|_{x \rightarrow \infty} = O(1),$$

$$u|_{y=0} = 0, \quad u|_{y=b} = 0.$$

$$\text{Ответ. } u = \frac{2u_0}{\pi} \int_0^\infty \left\{ 1 - \frac{\text{sh}(vy)}{\text{sh}(vb)} - \frac{\text{sh}(v(b-y))}{\text{sh}(vb)} \right\} \frac{\sin(vx)}{v} dv.$$

18. Решить задачу

$$\frac{\partial u}{\partial t} - a^2 \frac{\partial^2 u}{\partial x^2} = 0, \quad u|_{t=0} = \varphi(x), \quad -\infty < x < \infty, \quad t > 0.$$

$$\text{Ответ. } u(x, t) = \frac{1}{2a\sqrt{\pi t}} \int_{-\infty}^{+\infty} \varphi(\xi) e^{-\frac{(\xi-x)^2}{4a^2 t}} d\xi.$$

19. Пусть $u(x, t)$ — температура в некоторой одномерной области, причем начальная температура $u(x, 0) = f(x)$. Исследовать решения уравнения теплопроводности при различных начальных температурах: а) $f(x) = e^{-x^2}$ — гауссов начальный профиль; б) $f(x) = \theta(x+1) - \theta(x-1)$ — равномерное начальное распределение температуры на отрезке $[-1, 1]$, где $\theta(x)$ — единичная функция Хевисайда; в) $f(x) = (1-x)[\theta(x) - \theta(x-1)]$ — треугольный начальный профиль температуры.

$$\text{Ответ. } u(x, t) = \frac{1}{2c\sqrt{\pi t}} \int_{-\infty}^{+\infty} f(\xi) e^{-\frac{(x-\xi)^2}{4c^2 t}} d\xi;$$

$$\text{а) } u(x, t) = \frac{e^{-\frac{x^2}{4c^2t+1}}}{\sqrt{4c^2t+1}};$$

$$\text{б) } u(x, t) = \frac{1}{2} \Phi\left(\frac{x+1}{2c\sqrt{t}}\right) - \frac{1}{2} \Phi\left(\frac{x-1}{2c\sqrt{t}}\right);$$

$$\text{в) } u(x, t) = \frac{(1-x)}{2} \left\{ \Phi\left(\frac{x}{2c\sqrt{t}}\right) - \Phi\left(\frac{x-1}{2c\sqrt{t}}\right) \right\} - \frac{c\sqrt{t}}{\sqrt{\pi}} \left\{ e^{-\frac{x^2}{4c^2t}} - e^{-\frac{(x-1)^2}{4c^2t}} \right\}.$$

20. Решить плоскую стационарную задачу теории теплопроводности для квадранта (рис. 6.5), одна грань ($y = 0$) которого находится при температуре нуль, а другая покрыта тепловой изоляцией, за исключением участка $0 \leq y \leq b$, через который втекает поток тепла заданной плотности q . Получить распределение плотности теплового потока через грань $y = 0$.




Рис. 6.5. К задаче 20

Ответ. $u(x, y) = \frac{2q}{\pi k} \int_0^\infty \frac{1 - \cos(vb)}{v^2} e^{-vx} \sin(vy) dv$. где k — коэффициент теплопроводности; $q(x, 0) = \frac{q}{\pi} \ln\left(1 + \frac{b^2}{x^2}\right)$.

Указание. Требуется дать решение уравнения Лапласа в области $0 \leq x < \infty$,

$0 \leq y < \infty$ при граничных условиях $u|_{y=0} = 0, \frac{\partial u}{\partial x}|_{x=0} = \begin{cases} -\frac{q}{k}, & 0 < y < b, \\ 0, & 0 < y < \infty. \end{cases}$

7. Численные методы решения задач математической физики

Главное предназначение системы Maple — это выполнение символьных, аналитических преобразований в математике, и здесь она, бесспорно, занимает одно из лидирующих мест. Однако при решении практических задач, рано или поздно, всегда наступает момент, когда необходимо провести те или иные численные расчеты. Разработчики системы Maple, по-видимому, прекрасно это понимают, и с каждой новой версией эта система обогащается все новыми и новыми возможностями численного анализа.

Пока все же, при решении большинства серьезных численных задач, особенно для систем большого порядка, предпочтительнее использовать специальные программные средства численного анализа, такие, например, как MATLAB. Тем не менее, автору этой книги приходилось сталкиваться с численными задачами, решение которых удалось получить только с помощью Maple.

Как известно, реализация многих аналитических методов на цифровых компьютерах приводит к вычислительной неустойчивости большинства из них. Это связано с накоплением ошибок округления, возникающих при работе на множестве действительных чисел с ограниченным числом значащих цифр в мантиссе, которое реализуется на цифровых компьютерах. Простейший выход из подобной ситуации — увеличение количества значащих цифр, с помощью которых представляются числа на компьютере, но это — дорогостоящая операция, которая в ближайшее время, по всей видимости, решена не будет.

Выход из такого затруднительного положения уже сейчас видится в использовании для некоторых числовых расчетов систем аналитических вычислений, в которых проблемы с ограниченным количеством значащих цифр в мантиссе действительного числа не существует. Например, в системе Maple можно осуществлять расчеты на множестве действительных чисел, имеющих до 500 значащих цифр в мантиссе своего представления. Это, естественно, скажется на скорости вычислений и потребует использования более мощного компьютера. Однако для многих аналитических алгоритмов, учитывая их простоту, увеличение времени расчета не играет большой роли, так как порядок разрешающей алгебраической системы уравнений, во многих случаях, не превышает 100 – 150, позволяя получить удовлетворительный (с точки зрения точности получаемых решений) для практики результат.

В этой главе мы изучим возможности приближенного численного анализа задач математической физики, уже реализованные в Maple, а так же рассмотрим, как можно самому строить расчетные схемы для решения задачи Коши и краевой задачи.

Численное решение обыкновенных дифференциальных уравнений

Здесь мы рассмотрим приближенные численные методы решения задач для обыкновенных дифференциальных уравнений — задачи Коши и краевой задачи.

Задача Коши

Приближенные методы решения задачи Коши для обыкновенных дифференциальных уравнений можно разделить на две большие группы: аналитические, которые дают решение в виде функционального выражения, и численные, дающие решение задачи в виде таблицы числовых значений искомой функции при некотором наборе значений аргумента. Мы начнем рассмотрение с реализации классических численных методов решения задачи Коши. К таким методам можно отнести метод Эйлера и метод Рунге-Кутты.

Метод Эйлера

Рассмотрим решение задачи Коши для уравнения первого порядка методом Эйлера. Пусть требуется найти на промежутке $[x_0, x_0 + a]$ решение задачи Коши

$$\frac{dy}{dx} = f(x, y), \quad y(x_0) = y_0. \quad (7.1)$$

Промежуток $[x_0, x_0 + a]$ разбивают на некоторое число частичных промежутков точками x_0, x_2, \dots, x_n . При этом $\Delta x_n = x_{n+1} - x_n = h$ обычно считается постоянной величиной и называется шагом сетки, а точки x_i называются узлами сетки. При численном решении задачи ищут приближенные значения y_i для значений точного решения $y(x_i)$ в выбранных узлах на промежутке $[x_0, x_0 + a]$.

В методе Эйлера предполагается, что правая часть $f(x, y)$ дифференциального уравнения (7.1) есть величина постоянная на каждом из промежутков $[x_i, x_{i+1}]$, $i = 0, 1, 2, \dots, n$, равная $f(x_i, y_i)$.

Интегрируя при этом предположении дифференциальное уравнение (7.1) по промежутку $[x_i, x_{i+1}]$, мы приходим к расчетным формулам метода Эйлера

$$x_{i+1} = x_i + h, \quad y_{i+1} = y_i + hf(x_i, y_i). \quad (7.2)$$

Геометрически этот метод означает следующее: на промежутке $[x_i, x_{i+1}]$ считается постоянной производная $y'(x)$ искомого решения, то есть интегральная кривая заменяется отрезком прямой. Например, на отрезке $[x_0, x_1]$ угловой коэффициент прямой равен $f(x_0, y_0)$, то есть решение заменяется отрезком касательной в точке (x_0, y_0) к интегральной кривой.

Простейшая реализация этой схемы может быть представлена, например, с помощью следующей процедуры

```
> Euler := proc( ode, ic, domain, N )
    local h, i, t, y, F, L, X;
    t := lhs(domain);
```

```

y := op(0, lhs(ic));
h := ( op(2, rhs(domain)) - op(1, rhs(domain)) ) / N;
solve(ode, diff(y(t), t));
F := unapply(%, (t, y));
X := evalf( [ op(lhs(ic)), rhs(ic) ] );
L := X;
for i from 1 to N do
  X := X + [ h, h * F(op(X)) ];
  L := L, X;
end do;
return matrix(N+2, 2, [[t, y], L])
end proc:

```

Входными параметрами процедуры являются: переменная `ode`, содержащая дифференциальное уравнение; переменная `ic`, содержащая начальное условие в виде равенства $y(x_0) = y_0$; переменная `domain` — область изменения независимой переменной в виде равенства $x = x_0 \dots x_{\max}$; переменная `N` — количество шагов. Процедура возвращает матрицу значений (x, y) . Рассмотрим применение этой процедуры на примерах.

Задача 7.1. Решить с помощью процедуры `Euler` следующую задачу Коши

$$\frac{dy}{dx} = x + \cos\left(\frac{y}{\sqrt{7}}\right), \quad y(0.5) = 0.6$$

на отрезке $[0.5, 1.5]$.

Решение. Определяем уравнение, начальное условие и интервал, в котором ищется решение

```

> ode1 := diff( y(x), x ) = x + cos(y/sqrt(7.));
ic1 := y(0.5) = 0.6; domain := x = 0.5 .. 1.5;

```

$$ode1 := \frac{d}{dx} y(x) = x + \cos(0.3779644730 \ y)$$

$$ic1 := y(0.5) = 0.6$$

$$domain := x = 0.5 \dots 1.5$$

Решаем задачу с помощью команды

```

> Euler( ode1, ic1, domain, 10 );

```

x	y
0.5	0.6
0.6000000000	0.7474395730
0.7000000000	0.9034755714
0.8000000000	1.067701523
0.9000000000	1.239668671
1.000000000	1.418891042
1.100000000	1.604852043
1.200000000	1.797012454
1.300000000	1.994819573
1.400000000	2.197717210
1.500000000	2.405156140

Задача 7.2. Воспользовавшись методом Эйлера с количеством шагов $N = 2, 4, 8$, найти приближенное значение $y(2)$, являющееся решением задачи Коши

$$\frac{dy}{dt} \sin t + y(t) = 3, \quad y(1) = 2.$$

Решение. Определим задачу

```
> ode2 := diff( y(t), t ) * sin(t) + y(t) = 3;
```

```
ic2 := y(1)=2;
```

$$ode2 := \left(\frac{d}{dt} y(t) \right) \sin(t) + y(t) = 3$$

$$ic2 := y(1) = 2$$

Зададим координаты начальной точки и количество шагов в виде списка

```
> a2 := 1; b2 := 2; N2 := [ 2, 4, 8 ];
```

```
    a2 := 1
```

```
    b2 := 2
```

```
    N2 := [2, 4, 8]
```

Определим соответствующие величины шагов

```
> H2 := map( n->(b2-a2)/n, N2 );
```

$$H2 := \left[\frac{1}{2}, \frac{1}{4}, \frac{1}{8} \right]$$

Выполним теперь расчет для $N = 2$

```
> sol2 := Euler( ode2, ic2, t=a2..b2, N2[1] );
```

```
    Y2[1] := sol2[N2[1]+2,2];
```

```
        Y21 := 2.797608323
```

для $N = 4$

```
> sol2 := Euler( ode2, ic2, t=a2..b2, N2[2] ):
Y2[2] := sol2[N2[2]+2,2];
Y2_2 := 2.710599580
```

для $N = 8$

```
> sol2 := Euler( ode2, ic2, t=a2..b2, N2[3] ):
Y2[3] := sol2[N2[3]+2,2];
Y2_3 := 2.677425038
```

Результат удобно представить в виде таблицы. Для этого подключим пакет `linalg`

```
> with( linalg );
```

и подготовим следующую таблицу

```
> v0 := vector( [ 'h', 'N', 'Y(2)' ] ):
v1 := vector( [ H2[1], N2[1], Y2[1] ] ):
v2 := vector( [ H2[2], N2[2], Y2[2] ] ):
v3 := vector( [ H2[3], N2[3], Y2[3] ] ):
stackmatrix( v0, v1, v2, v3 );
```

$$\begin{bmatrix} h & N & Y(2) \\ \frac{1}{2} & 2 & 2.797608323 \\ \frac{1}{4} & 4 & 2.710599580 \\ \frac{1}{8} & 8 & 2.677425038 \end{bmatrix}$$

Метод Эйлера (7.2) очень прост, но основной его недостаток — быстрое нарастание погрешности расчета. Заменяя отрезок истинной интегральной кривой отрезком прямой, касательной к ней, мы на протяжении интервала h несколько отдаляемся от кривой. Вычисляя в новой точке значение производной, мы получаем некоторую погрешность, обусловленную различием расчетной и истинной ординат интегральной кривой. К этой погрешности добавляется новая погрешность от замены кривой прямолинейным отрезком и т. д. В результате точность расчета оказывается невысокой. Уменьшение величины интервалов увеличивает точность расчетов, но незначительно; можно приближенно считать, что погрешность пропорциональна величине интервала h . Этот метод является методом первого порядка точности.

Метод Эйлера можно усовершенствовать различными способами. Одна из схем усовершенствованного метода Эйлера для решения задачи (7.1) имеет вид

$$x_{n+1} = x_n + h,$$

$$\begin{aligned}
y_{n+1}^* &= y_n + hf(x_n, y_n), \\
y_{n+1} &= y_n + \frac{h}{2} [f(x_n, y_n) + f(x_{n+1}, y_{n+1}^*)],
\end{aligned}
\tag{7.3}$$

где h — величина шага. Будем называть метод (7.3) *усовершенствованным методом Эйлера*, иногда его называют *трапецидальным методом*.

Можно предложить и другую схему усовершенствования

$$\begin{aligned}
x_{n+1} &= x_n + h, \\
y_{n+\frac{1}{2}} &= y_n + \frac{h}{2} f(x_n, y_n), \\
y_{n+1} &= y_n + hf\left(x_n + \frac{h}{2}, y_{n+\frac{1}{2}}\right).
\end{aligned}
\tag{7.4}$$

Будем называть метод (7.4) *модифицированным методом Эйлера*.

Простейшая реализация схемы модифицированного метода Эйлера может быть осуществлена, например, с помощью следующей процедуры

```

> ImprEuler := proc( ode, ic, domain, N )
  local h, i, t, y, F, L, X, X1, X2;
  t:=lhs(domain);
  y:=op(0, lhs(ic));
  h:=(op(2, rhs(domain))-op(1, rhs(domain)))/N;
  solve(ode, diff(y(t), t)):
  F:=unapply(%, (t, y));
  X:=evalf([op(lhs(ic)), rhs(ic)]);
  L:=X;
  for i from 1 to N do
    X1:=X + [h, h*F(op(X))];
    X2:=X + [h, h*F(op(X1))];
    X:=(X1 + X2)/2;
    L:=L, X;
  end do;
  return matrix(N+2, 2, [[t, y], L])
end proc:

```

Здесь входные параметры те же, что и в процедуре Euler.

Для сравнения рассмотрим решения задач 7.1 и 7.2 модифицированным методом Эйлера. Решение задачи 7.1:

```
> ImprEuler( ode1, ic1, domain, 10 );
```

x	y
0.5	0.6
0.6000000000	0.7517377857
0.7000000000	0.9118190782
0.8000000000	1.079801236
0.9000000000	1.255203548
1.0000000000	1.437513741
1.1000000000	1.626195858
1.2000000000	1.820699250
1.3000000000	2.020468379
1.4000000000	2.224953054
1.5000000000	2.433618715

Решение задачи 7.2:

```
> sol2 := ImprEuler( ode2, ic2, t=a2..b2, N2[1] );
```

```
Y2[1] := sol2[N2[1]+2,2];
```

```
Y21 := 2.631918672
```

```
> sol2 := ImprEuler( ode2, ic2, t=a2..b2, N2[2] );
```

```
Y2[2] := sol2[N2[2]+2,2];
```

```
Y22 := 2.646365728
```

```
> sol2 := ImprEuler( ode2, ic2, t=a2..b2, N2[3] );
```

```
Y2[3] := sol2[N2[3]+2,2];
```

```
Y23 := 2.648639612
```

```
> v0 := vector( [ 'h', 'N', 'Y(2)' ] );
```

```
v1 := vector( [ H2[1], N2[1], Y2[1] ] );
```

```
v2 := vector( [ H2[2], N2[2], Y2[2] ] );
```

```
v3 := vector( [ H2[3], N2[3], Y2[3] ] );
```

```
stackmatrix( v0, v1, v2, v3 );
```

h	N	$Y(2)$
$\frac{1}{2}$	2	2.631918672
$\frac{1}{4}$	4	2.646365728
$\frac{1}{8}$	8	2.648639612

Мы убедились на рассмотренных задачах, что не представляет труда запрограммировать тот или иной численный метод для решения обыкновенного дифференциального уравнения. На самом деле необходимость в программировании численных методов решения задач для обыкновенных дифференциальных уравнений в Maple отпадает. Как мы отмечали ранее, можно использовать команду `dsolve` для получения приближенного решения задачи Коши для обыкновенного дифференциального уравнения и для системы таких уравнений.

Функция `dsolve/numeric` находит численное решение обыкновенных дифференциальных уравнений. Вызов этой функции осуществляется в следующих вариантах

```
dsolve(odesys, numeric, vars, options);
dsolve(numeric, vars, procopts, options);
```

Параметры:

- `odesys` — набор или список; обыкновенное дифференциальное уравнение (или система уравнений) и начальные / граничные условия;
- `numeric` — ключевое слово, указывающее функции `dsolve` на необходимость искать численное решение;
- `vars` — (необязательный) зависимая переменная или набор или список зависимых переменных для `odesys`;
- `procopts` — (требуется, если `odesys` отсутствует) опции, которые обуславливают процедуру определенной системы;
- `options` — (необязательный) уравнения вида `keyword = value`.

Вызов функции `dsolve` с опцией `numeric` или `type=numeric` заставляет `dsolve` находить численное решение системы обыкновенных дифференциальных уравнений или одного уравнения. Если необязательное уравнение `method=numericmethod` указывается (где `numericmethod` — одно из значений `rkf45`, `rosenbrock`, `bvp`, `dverk78`, `lsode`, `gear`, `taylorseries`, или `classical`), функция `dsolve` использует указанный метод для получения численного решения.

Функцией `dsolve` могут быть численно решены как начальные, так и краевые задачи. По умолчанию, результат работы процедуры представлен так, что он может использоваться, для получения значения решения, если задано значение независимой переменной, причем, для решения задач Коши используется метод Рунге-Кутты пятого порядка точности, а для краевых задач используется методика конечных разностей с экстраполяцией Ричардсона.

Методы решения задач Коши, за исключением `taylorseries`, имеют возможность работы с комплексными функциями от действительной независимой переменной. Методы решения краевых задач пока не могут использоваться для комплексных проблем, и требуют преобразования системы к действительной системе перед обращением к `dsolve`.

В случае использования методов `rkf45`, `rosenbrock`, `lsode`, и `gear`, пользователь может устанавливать для функции `dsolve` переменную `infolevel=2` так, чтобы предоставлялась информация относительно ошибки в последних значениях. Это полезно когда мы имеем дело с асимптотическим поведением решения.

Высокоуровневые опции, наиболее общие как для задачи Коши, так и для краевой задачи, следующие:

- `'output'` = ключевое слово или массив;
- `'stiff'` = булева переменная;
- `'stop_cond'` = list;
- `'range'` = numeric .. numeric;
- `'abserr'` = numeric;
- `'relerr'` = numeric.

Ключевое слово `output` может принимать значения `procedurelist`, `listprocedure`, или `piecewise` или это может быть `array` или `Array`, который задает значения независимой переменной, в которых желательно получить значения решения.

Заданное по умолчанию ключевое слово `procedurelist` определяет результат работы `dsolve` как процедура. Эта процедура принимает значение независимой переменной как параметр, и возвращает список значений решения в форме равенства `variable=value`, где левые части — имена независимой переменной, зависимой переменной (или переменных) и их производных (для уравнений более высоких порядков), и правые части — соответствующие вычисленные значения решения.

Ключевое слово `listprocedure` определяет результат как список уравнений в форме `variable=procedure`, где левые части — имена независимой переменной, зависимой переменной (или переменных) и производных, и правые части — процедуры, которые могут использоваться, чтобы вычислить соответствующие компоненты решения. Эта выходная форма наиболее полезна, когда возвращенная процедура должна использоваться позже, например, с функцией `fsolve`.

Ключевое слово `piecewise` доступно только для нежестких и жестких методов решения задачи Коши, заданных по умолчанию (`rkf45`, `rosenbrock`), и дает результат как список уравнений в форме `variable=pwfunc`, где левые части — имена независимой переменной, зависимой переменной (или переменных) и производных, и правые части — кусочные функции, описывающие соответствующие компоненты решения. Эти кусочные функции получены методом `interpolants` для каждого шага вычисления. Использование этой выходной формы требует, чтобы использовал-

ся параметр диапазона для сообщения функции `dsolve/numeric` желательного диапазон кусочной функции.

Когда задается опция `array` (это должен быть вектор значений с плавающей точкой, определяющий значения независимой переменной, в которых желательно получить значения решения) — возвращается 2×1 – матрица (вместо процедуры решения). Причем `[1,1]`-вход — массив, содержащий название независимой переменной и имена зависимой переменной (переменных) и производных. А `[2,1]`-вход — матрица, первый столбец которой — копия выходного вектора, то есть значения независимой переменной; другие столбцы — значения зависимой переменной (переменных) и производных, соответствующих вектору имен в `[1,1]`. Строка i этой матрицы — вектор, содержащий, и значение независимой переменной, и значения зависимых переменных, оцененных в элементе i выходного массива `output`.

Если используется опция `Array`, то результат получается тот же самый что и при использовании опции `array`, за исключением того, что используется более новый тип (`datatypes`) `Array` и `Matrix` на выходе. Для большого количества выходных точек и (или) компонентов решения, выходные данные не могут быть прямо видимы, потому что имеется заданный по умолчанию размер, в котором отображается представление `Array` или `Matrix`.

Булева переменная `stiff` используется только для решения задачи Коши. Установка `stiff=true` без выбора метода сообщает функции `dsolve`, что заданный по умолчанию жесткий метод (`rosenbrock`) должен использоваться вместо заданного по умолчанию нежесткого метода (`rkf45`). Когда метод к тому же определен, выполняется проверка на непротиворечивость, что метод соответствует значению `stiff`.

Параметр `stop_cond` доступен только для действительной задачи Коши, использующей нежесткие и жесткие, заданные по умолчанию, методы `rkf45` и `rosenbrock`, и при этом принимается список критериев остановки. Это список условий, которые проверяются в процессе численного интегрирования обыкновенных дифференциальных уравнений. Когда встречается одно из таких условий, интегрирование прекращается, и текущая точка на выходе выдается как решение.

Значения `range` сообщают функции `dsolve` диапазон независимой переменной, по которой желательно получить значения решения. Для задачи Коши, эта опция используется только нежесткими и жесткими заданными по умолчанию методами (`rkf45`, `rosenbrock`), и имеет две цели для процедурного типа `output`. Если диапазон используется, то запрос к функции `dsolve` вычислит решение по заданному диапазону перед возвращением, сохраняя это решение для возможных более поздних запросов к возвращенной процедуре, которая тогда вычислит возвращаемые значения с помощью интерполяции.

При получении численного решения для задачи, которая имеет большие области, где решение изменяется медленно, и малые области где, решение изменяется быстро, использование диапазона совместно с опцией `refine` функции `odeplot` способствует лучшему составлению графика решения. Если диапазон не используется, то запрос к функции `dsolve` возвратит процедуру, которая не сохраняет вычисленное решение (вычисляет решение всякий раз в требуемой точке).

Числовое значение `abserr` задает предельный допуск абсолютной погрешности для шага в случае задачи Коши, и критерий предельной ошибки между вычисленным решением и точным решением в случае краевой задачи. Эта опция поддержана всеми методами кроме `classical`.

Числовое значение `relerr` задает предельный допуск относительной ошибки для шага в случае задачи Коши. Эта опция работает вместе с опцией `abserr` и поддерживается всеми методами кроме `classical`, `taylorseries`, и `bvp`.

На этом мы закончим краткое описание функции `dsolve/numeric`, отсылая читателя за более подробной информацией к справочным страницам Maple.

Рассмотрим на примерах применение функции `dsolve/numeric`.

Для иллюстрации снова рассмотрим задачи 7.1 и 7.2.

Если мы хотим получить явную таблицу приближенного решения, то подготовим список значений независимой переменной, в которых мы хотим получить решение

```
> odel; ic1;
```

$$\frac{d}{dx} y(x) = x + \cos(0.3779644730 \ y(x))$$

$$y(0.5) = 0.6$$

```
> x0 := 0.5; h1 := 0.1; N1 := 10;
```

```
    x0 := 0.5
```

```
    h1 := 0.1
```

```
    N1 := 10
```

```
> x_list := vector( N1+1, i -> x0 + (i-1)*h1 );
```

```
    x_list := [0.5, 0.6, 0.7, 0.8, 0.9, 1.0, 1.1, 1.2, 1.3, 1.4, 1.5]
```

Теперь таблицу приближенных значений искомого решения найдем методом Эйлера с помощью функции `dsolve/numeric`


```
> dsolve( { odel, ic1 }, y(x), type=numeric,
```

```
    method=classical, stepsize=h1, output=x_list );
```

[x, y(x)]	
0.5	0.6
0.6	0.747439572962070864
0.7	0.903475571397945321
0.8	1.06770152276314744
0.9	1.23966867119627011
1.0	1.41889104244383812
1.1	1.60485204361813372
1.2	1.79701245411977606
1.3	1.99481957329573478
1.4	2.19771721012666308
1.5	2.40515613980256093

Если результат необходимо отобразить на графике, можно использовать команды `dsolve` и `odeplot`, как это показано ниже. Для того чтобы воспользоваться командой `odeplot`, надо подключить пакет `plots`

```
> with( plots );
sol := dsolve( { odel, ic1 }, y(x), type=numeric,
               method=classical, stepsize=h1 );
picE:=odeplot( sol, [x,y(x)], 0.5..1.5, color=black,
               legend=`метод Эйлера`, style=point );
display({picE});
```



Задача 7.2 также легко решается с помощью команды `dsolve`

```
> ode2; ic2;
```

$$\left(\frac{d}{dt} y(t) \right) \sin(t) + y(t) = 3$$

$$y(1) = 2$$

```

> sol2 := dsolve( { ode2, ic2 }, y(t), type=numeric,
method=classical, stepsize=H2[1] ):
Y2[1] := eval( y(t), sol2(2) );
          Y21 := 2.79760832314891060

> sol2 := dsolve( { ode2, ic2 }, y(t), type=numeric,
method=classical, stepsize=H2[2] ):
Y2[2] := eval( y(t), sol2(2) );
          Y22 := 2.71059958025902015

> sol2 := dsolve( { ode2, ic2 }, y(t), type=numeric,
method=classical, stepsize=H2[3] ):
Y2[3] := eval( y(t), sol2(2) );
          Y23 := 2.67742503870978908

> v0 := vector( [ 'h', 'N', 'Y(2)' ] ):
v1 := vector( [ H2[1], N2[1], Y2[1] ] ):
v2 := vector( [ H2[2], N2[2], Y2[2] ] ):
v3 := vector( [ H2[3], N2[3], Y2[3] ] ):
stackmatrix( v0, v1, v2, v3 );

```

$$\begin{bmatrix} h & N & Y(2) \\ \frac{1}{2} & 2 & 2.797608323 \\ \frac{1}{4} & 4 & 2.710599580 \\ \frac{1}{8} & 8 & 2.677425039 \end{bmatrix}$$

Рассмотрим решение этих же задач модифицированным методом Эйлера с помощью команды dsolve. Задача 7.1:


```

> dsolve( { ode1, ic1 }, y(x), type=numeric,
method=classical[heunform], stepsize=h1,
output=x_list );

```


[x, y(x)]	
0.5	0.6
0.6	0.751737785698972649
0.7	0.911819078174192610
0.8	1.07980123658405014
0.9	1.25520354816513224
1.0	1.43751374120181685
1.1	1.62619585832199598
1.2	1.82069925079892392
1.3	2.02046837952947689
1.4	2.22495305421199241
1.5	2.43361871556365994

```
> sol := dsolve( { odel, ic1 }, y(x), type=numeric,
                 method=classical[heunform], stepsize=h1 ):
pic_mod:=odeplot( sol, [x,y(x)], 0.5..1.5, color=black,
                 legend=`модифицированный метод Эйлера` ):
display({pic_mod,picE});
```



Задача 7.2:

```
> sol2 := dsolve( { ode2, ic2 }, y(t), type=numeric,
                 method=classical[heunform],
                 stepsize=H2[1] ):
Y2[1] := eval( y(t), sol2(2) );
Y2_1 := 2.63191867242761024

> sol2 := dsolve( { ode2, ic2 }, y(t), type=numeric,
```

```

method=classical[heunform],
stepsize=H2[2] ):
Y2[2] := eval( y(t), sol2(2) );
Y2_2 := 2.64636572746602816
> sol2 := dsolve( { ode2, ic2 }, y(t), type=numeric,
method=classical[heunform],
stepsize=H2[3] ):
Y2[3] := eval( y(t), sol2(2) );
Y2_3 := 2.64863961122298264
> v0 := vector( [ 'h', 'N', 'Y(2)' ] ):
v1 := vector( [ H2[1], N2[1], Y2[1] ] ):
v2 := vector( [ H2[2], N2[2], Y2[2] ] ):
v3 := vector( [ H2[3], N2[3], Y2[3] ] ):
stackmatrix( v0, v1, v2, v3 );

```

$$\begin{bmatrix} h & N & Y(2) \\ \frac{1}{2} & 2 & 2.631918672 \\ \frac{1}{4} & 4 & 2.646365727 \\ \frac{1}{8} & 8 & 2.648639611 \end{bmatrix}$$

Метод Рунге-Кутты

Рассмотрим еще одну популярную группу методов решения задачи Коши (7.1) для обыкновенного дифференциального уравнения — методы Рунге-Кутты.

Представим искомое решение $y = y(x)$ задачи (7.1) в окрестности каждой точки $x = x_n$, $n = 0, 1, 2, \dots$ рядом Тейлора и вычислим $y(x_{n+1})$. Тогда, если принять обозначение $x_{n+1} - x_n = h$, получим следующее равенство

$$y(x_{n+1}) = y(x_n) + h \left. \frac{dy}{dx} \right|_{x=x_n} + \frac{h^2}{2!} \left. \frac{d^2y}{dx^2} \right|_{x=x_n} + \frac{h^3}{3!} \left. \frac{d^3y}{dx^3} \right|_{x=x_n} + \frac{h^4}{4!} \left. \frac{d^4y}{dx^4} \right|_{x=x_n} + \dots$$

В зависимости от числа оставляемых членов в последней формуле, получаем вычислительную схему с той или иной степенью точности. Заметим, что если удерживать в разложении члены до первого порядка включительно, то будем иметь вычислительную схему метода Эйлера. Таким образом, метод Эйлера становится частным случаем метода Рунге-Кутты.

Одним из широко распространенных методов является метод Рунге-Кутты второго порядка, вычисления по которому реализуются по формулам

$$\begin{aligned} k_1 &= f(x_n, y_n), \\ k_2 &= f(x_n + h, y_n + hk_1), \\ y_{n+1} &= y_n + \frac{h}{2}(k_1 + k_2). \end{aligned} \quad (7.5)$$

Геометрический смысл этих формул состоит в том, что через точку (x_n, y_n) проводится не касательная к интегральной кривой, а прямая с угловым коэффициентом, равным среднему арифметическому угловых коэффициентов касательных, которые строятся в методе Эйлера для двух последовательных шагов. В качестве решения берется ордината точки пересечения этой прямой с вертикалью $x = x_{n+1}$.

Приведем еще одну популярную вычислительную схему метода Рунге-Кутты четвертого порядка

$$\begin{aligned} k_1 &= f(x_n, y_n), \\ k_2 &= f\left(x_n + \frac{h}{2}, y_n + \frac{h}{2}k_1\right), \\ k_3 &= f\left(x_n + \frac{h}{2}, y_n + \frac{h}{2}k_2\right), \\ k_4 &= f(x_n + h, y_n + hk_3), \\ y_{n+1} &= y_n + \frac{h}{6}(k_1 + 2k_2 + 2k_3 + k_4). \end{aligned}$$

Эти схемы легко программируются. Например, программа, реализующая схему (7.5), может выглядеть так

```
> RungeKutta2 := proc( ode, ic, domain, N )
  local f, h, i, t, y, F, L, X, X1, X2;
  t := lhs(domain);
  y := op(0, lhs(ic));
  h := ( op(2, rhs(domain)) - op(1, rhs(domain)) ) / N;
  solve( ode, diff(y(t), t) ):
  F := unapply( %, (t, y) );
  X := evalf( [ op(lhs(ic)), rhs(ic) ] );
  L := X;
  for i from 1 to N do
```

```

f := F(op(X));
X1 := X + [ h, h*f ];
X2 := X + [ h, h*F(op(X1)) ];
X := (X1 + X2)/2;
L := L, X;
end do;
return matrix(N+2,2,[[t,y],L])
end proc;

```

В качестве иллюстрации использования этой программы снова рассмотрим задачи 7.1 и 7.2.

Решение задачи 7.1:

```

> ode1 := diff( y(x), x ) = x+cos(y(x)/sqrt(7.));
ic1 := y(0.5)=0.6; domain:=x=0.5..1.5;

```

$$ode1 := \frac{d}{dx} y(x) = x + \cos(0.3779644730 y(x))$$

$$ic1 := y(0.5) = 0.6$$

$$domain := x = 0.5 .. 1.5$$

```

> RungeKutta2( ode1, ic1, x=0.5..1.5, 10 );

```

x	y
0.5	0.6
0.6000000000	0.7517377857
0.7000000000	0.9118190782
0.8000000000	1.079801236
0.9000000000	1.255203548
1.000000000	1.437513741
1.100000000	1.626195858
1.200000000	1.820699250
1.300000000	2.020468379
1.400000000	2.224953054
1.500000000	2.433618715

Решение задачи 7.2:

```

> ode2 := diff( y(t), t ) * sin(t) + y(t) = 3;
ic2 := y(1)=2;

```

$$ode2 := \left(\frac{d}{dt} y(t) \right) \sin(t) + y(t) = 3$$

```

        ic2 := y(1) = 2
> a2 := 1; b2 := 2; N2 := [ 2, 4, 8 ];
        a2 := 1
        b2 := 2
        N2 := [2, 4, 8]
> H2 := map( n->(b2-a2)/n, N2 );
        H2 := [1/2, 1/4, 1/8]
> sol2 := RungeKutta2( ode2, ic2, t=a2..b2, N2[1] ):
        Y2[1] := sol2[N2[1]+2,2];
        Y2_1 := 2.631918672
> sol2 := RungeKutta2( ode2, ic2, t=a2..b2, N2[2] ):
        Y2[2] := sol2[N2[2]+2,2];
        Y2_2 := 2.646365728
> sol2 := RungeKutta2( ode2, ic2, t=a2..b2, N2[3] ):
        Y2[3] := sol2[N2[3]+2,2];
        Y2_3 := 2.648639612
> v0 := vector( [ 'h', 'N', 'Y(2)' ] ):
        v1 := vector( [ H2[1], N2[1], Y2[1] ] ):
        v2 := vector( [ H2[2], N2[2], Y2[2] ] ):
        v3 := vector( [ H2[3], N2[3], Y2[3] ] ):
        stackmatrix( v0, v1, v2, v3 );

```

$$\begin{bmatrix} h & N & Y(2) \\ \frac{1}{2} & 2 & 2.631918672 \\ \frac{1}{4} & 4 & 2.646365728 \\ \frac{1}{8} & 8 & 2.648639612 \end{bmatrix}$$

Решение этих задач можно осуществить и с помощью команды `dsolve`, указав опцию `type=numeric` и метод решения `method=classical[rk2]`.

Краевая задача

Перейдем теперь к рассмотрению численного решения краевых задач для обыкновенных дифференциальных уравнений.

Одним из широко применяемых методов численного решения краевых задач является метод сведения краевой задачи к задаче Коши с последующим применением при решении последней известных численных методов, например, методов Рунге-Кутты. Альтернативой такому подходу является метод конечных разностей.

Метод пристрелки

Рассмотрим один из вариантов метода сведения краевой задачи к задаче Коши, известный, как *метод пристрелки*.

В методе пристрелки выбирается недостающее (незаданное) начальное условие в начальной точке интервала, а потом решается задача Коши — численно интегрируется дифференциальное уравнение до конечной точки интервала. Затем, сравнивая *вычисленное* значение зависимой переменной в конечной точке с *заданным*, проверяют правильность выбора недостающего начального условия. Если эти значения не совпадают, то выбирают другое значение недостающего начального условия и процесс повторяется до тех пор, пока разность между вычисленным и заданным условиями в конечной точке не станет достаточно малой. При таком подходе естественно возникает вопрос: существует ли систематический способ нахождения каждого последующего значения недостающего начального условия?

В качестве такого способа можно использовать, например, метод Ньютона, метод секущих или еще какой-нибудь метод, основанный на методе решения нелинейных алгебраических уравнений. При таком подходе сохраняется нелинейная форма дифференциального уравнения, а недостающее значение производной находится систематически одним из перечисленных методов.

Метод пристрелки является общим методом и может использоваться для решения нелинейных задач. Мы проиллюстрируем применение этого метода для решения двухточечной краевой задачи для линейного дифференциального уравнения следующего вида

$$\frac{d^2y}{dx^2} + p(x)\frac{dy}{dx} + q(x)y = f(x), \quad y(a) = r, \quad y(b) = s. \quad (7.6)$$

Соответствующая задача Коши, которая лежит в основе метода пристрелки, имеет вид

$$\frac{d^2y}{dx^2} + p(x)\frac{dy}{dx} + q(x)y = f(x), \quad y(a) = r, \quad \left. \frac{dy}{dx} \right|_{x=b} = \alpha, \quad (7.7)$$

где α — неизвестный параметр.

Задаемся конкретным значением параметра α и решаем задачу (7.7); находим значение функции на правом конце интервала, которое, естественно, зависит от параметра α : $y(b, \alpha)$. Наша основная цель состоит в подборе такого значения α , чтобы выполнялось условие $y(b, \alpha) = s$. Для достижения этой цели разработаем процедуру, решающую задачу Коши и определяющую значение зависимой переменной на правом конце интервала. Это может быть, например, такая процедура

```
> Yb := proc ( ivp, fn, b )
```

```

local sol;
sol := dsolve( ivp, fn, type=numeric );
eval( fn, sol(b) )
end proc:

```

Входными параметрами процедуры являются задача Коши, неизвестная функция и правая граничная точка интервала, на котором ищется решение краевой задачи.

Для определения параметра α из уравнения $y(b, \alpha) - s = 0$ будем использовать так называемый *метод секущих* или *метод линейной интерполяции*. Рассмотрим все вычисления на конкретном примере.

Задача 7.3. Решить следующую двухточечную краевую задачу

$$\frac{d^2y}{dx^2} + 3\frac{dy}{dx} + 7y = 2e^{-x} - 5, \quad y(0) = 2, \quad y(1) = 5.$$

Решение. Будем решать эту задачу численно методом пристрелки. Определим задачу в Maple

```

> ode1:=diff(y(x), x$2)+3*diff(y(x), x)+7*y(x)=2*exp(-x)-5;
bc1:=y(0)=2, y(1)=-5;

```

$$ode1 := \left(\frac{d^2}{dx^2} y(x) \right) + 3 \left(\frac{d}{dx} y(x) \right) + 7 y(x) = 2 e^{(-x)} - 5$$

$$bc1 := y(0) = 2, y(1) = -5$$

Задаем граничные точки и формулируем начальные условия соответствующей задачи Коши

```

> a := 0; b := 1;
a := 0
b := 1
> ic1 := bc1[1], D(y)(a)=alpha;
ic1 := y(0) = 2, D(y)(0) = alpha

```

Здесь параметр α пока неизвестен. Этот параметр должен быть выбран из условия

```

> bound := lhs(bc1[2]) - rhs(bc1[2]);
> bound = 0;
y(1) + 5 = 0

```

Для определения параметра α используем метод секущих, реализация которого может быть осуществлена, например, так

```

> a0:=0;
> y0:=

```

```

eval (bound, y(b)=Yb (eval ({odel, ic1}, alpha=a0), y(x), b));
a1:=1:
y1:=
eval (bound, y(b)=Yb (eval ({odel, ic1}, alpha=a1), y(x), b));
while fnormal(y1-y0)<>0 do
z:=solve(y1 = (y1-y0)/(a1-a0)*(a1-x), x);
a0, a1 := a1, z;
y0, y1:=
y1, eval (bound, y(b)=Yb (eval ({odel, ic1}, alpha=z), y(x), b));
end do:
alpha_opt1 := z;

y0 := 4.462791110
y1 := 4.546784868
alpha_opt1 := -53.13240977

```

Решение задачи методом пристрелки — есть решение задачи Коши с уточненным параметром α :

```

> shoot_sol1:=
evalf (dsolve (eval ({odel, ic1}, alpha=alpha_opt1), y(x)));
shoot_sol1 := y(x) = -22.60248831 e(-1.500000000 x) sin(2.179449472 x)
+ 2.314285714 e(-1.500000000 x) cos(2.179449472 x)
+ 0.02857142857 (14. - 25. ex) e(-1. x)

```

Не удивительно, что в этом простом примере Maple оказался в состоянии найти точное решение задачи. В общем случае задача Коши решается численно.

Для сравнения найдем точное решение исходной краевой задачи, которое тоже легко получается в Maple

```

> exact_sol1:=
evalf (combine (dsolve ({odel, bcl}, y(x))));
exact_sol1 := y(x) = -24.21534749 e(-1.500000000 x) sin(2.179449472 x)
- 0.7142857145
- 2.820858789 e(-1.500000000 x) sin(2.179449472 x - 2.179449472 )
+ 0.3999999998 e(-1. x)

```

Для наглядности отобразим эти решения на одном графике

```

> rex:=


```



```

plot(rhs(exact_sol1), x=0..1, color=BLACK,
legend=`точное решение`):
psh:=
plot(rhs(shoot_sol1), x=0..1, color=BLACK, style=point,
legend=`метод пристрелки`):
plots[display]({pex, psh});

```



Из приведенных графиков видно хорошее совпадение решений. Отметим, что значение решения, полученное на правой границе методом пристрелки, есть

```

> evalf( eval( rhs(shoot_sol1), x=1 ) );
-5.000000639

```

Это также свидетельствует о хорошем согласии между решением, полученным методом пристрелки, и точным решением.

Рассмотрим еще один пример.

Задача 7.4. Решить двухточечную краевую задачу

$$x^2 \frac{d^2 y}{dx^2} + 2x \frac{dy}{dx} + 4y = -x^3, \quad y(1) = 7, \quad y(4) = -1.$$

Решение. Определяем задачу в Maple

```

> ode2 := x^2 * diff(y(x), x$2) + 2 * x * diff(y(x), x) + 4 * y(x) = -x^3;
bc2 := y(1) = 7, y(4) = -1;

```

$$ode2 := x^2 \left(\frac{d^2}{dx^2} y(x) \right) + 2x \left(\frac{d}{dx} y(x) \right) + 4y(x) = -x^3$$

$$bc2 := y(1) = 7, y(4) = -1$$

Формируем начальные условия для задачи Коши

```

> a := 1; b := 4;

a := 1
b := 4

```

```
> ic2 := bc2[1], D(y)(a)=alpha;
      ic2 := y(1) = 7, D(y)(1) = alpha
```

Параметр α определяем из условия:

```
> bound := lhs(bc2[2]) - rhs(bc2[2]);
> bound = 0;
      y(4) + 1 = 0
```

методом секущих

```
> a0 := 0:
y0:=
eval(bound, y(b)=Yb(eval({ode2, ic2}, alpha=a0), y(x), b)):
a1 := 1:
y1:=
eval(bound, y(b)=Yb(eval({ode2, ic2}, alpha=a1), y(x), b)):
while fnormal(y1-y0)<>0 do
  z := solve( y1 = (y1-y0)/(a1-a0)*(a1-x), x );
  a0, a1 := a1, z;
  y0, y1:=y1,
  eval(bound, y(b)=Yb(eval({ode2, ic2}, alpha=z), y(x), b));
end do:
alpha_opt2 := z;
      alpha_opt2 := 50.42074860
```

Решаем теперь окончательно задачу Коши и, таким образом, получаем приближенное решение исходной краевой задачи

```
> shoot_sol2:=
evalf(dsolve(eval({ode2, ic2}, alpha=alpha_opt2), y(x)));
      shoot_sol2 := y(x) =  $\frac{27.95751686 \sin(1.936491673 \ln(x))}{\sqrt{x}}$ 
      +  $\frac{7.062500000 \cos(1.936491673 \ln(x))}{\sqrt{x}}$  - 0.06250000000 x3
```

Здесь тоже решение задачи Коши удалось получить точно.


Для сравнения найдем точное решение исходной краевой задачи и отобразим полученные решения на одном графике

```
> exact_sol2:=evalf(combine(dsolve({ode2, bc2}, y(x))));
```

```

exact_sol2 := y(x) = 0.1416274967 (
    113. sin(-3.872983346 ln(sqrt(x)) + 2.684547487 )
    + 96. sin(3.872983346 ln(sqrt(x))) - 0.4412984869 x^(7/2) ) / sqrt(x)
> pex2:=plot(rhs(exact_sol2), x=1..4, color=BLACK,
    legend=`точное решение`):
psh2:=plot(rhs(shoot_sol2), x=1..4, color=BLACK,
    style=point, legend=`метод пристрелки`):
plots[display]({pex2, psh2});

```



Из приведенных графиков видно хорошее совпадение решений. Значение решения, полученное на правой границе методом пристрелки, есть

```

> evalf(eval(rhs(shoot_sol2), x=4));
-0.999998872

```

что также свидетельствует о хорошем согласии между решением, полученным методом пристрелки, и точным решением.

Рассмотрим еще один пример нелинейной задачи.

Задача 7.5. Решить следующую двухточечную краевую задачу для нелинейного уравнения

$$\frac{d^2 y}{dx^2} = x \frac{dy}{dx} - y^2, \quad y(0) = 2, \quad y(1) = -5.$$

Решение:

```

> ode3:=diff(y(x), x$2)=x*diff(y(x), x)-y(x)^2;
bc3:=y(0)=2, y(1)=-5;

```

$$ode3 := \frac{d^2}{dx^2} y(x) = x \left(\frac{d}{dx} y(x) \right) - y(x)^2$$

$$bc3 := y(0) = 2, y(1) = -5$$

```

> a := 0; b := 1;

```

$$a := 0$$

$$b := 1$$

```

> ic3 := bc3[1], D(y)(a)=alpha;
          ic3 := y(0) = 2, D(y)(0) =  $\alpha$ 
> bc := lhs(bc3[2]) - rhs(bc3[2]);
> bc = 0;
          bc := y(1) + 5
          y(1) + 5 = 0
> a0 := 0:
y0:=eval(bc, y(b)=Yb(eval({ode3, ic3}, alpha=a0), y(x), b)):
a1 := 1:
y1:=eval(bc, y(b)=Yb(eval({ode3, ic3}, alpha=a1), y(x), b)):
while fnormal(y1-y0)<>0 do
z := solve( y1 = (y1-y0)/(a1-a0)*(a1-x), x );
a0, a1 := a1, z;
y0, y1:=y1,
eval(bc, y(b)=Yb(eval({ode3, ic3}, alpha=z), y(x), b));
end do:
alpha_opt3 := z;
          alpha_opt3 := -4.861659299

```

Здесь решение соответствующей задачи Коши получим численно

```

> shoot_sol3 :=
dsolve( eval( {ode3, ic3}, alpha=alpha_opt3 ), y(x),
numeric, output=listprocedure );
shoot_sol3 := [ x = (proc(x) ... end proc ), y(x) = (proc(x) ... end proc ),
 $\frac{d}{dx}y(x) = (\text{proc}(x) \dots \text{end proc}) ]$ 
> y_shoot := eval(y(x), shoot_sol3);
          y_shoot := proc(x) ... end proc

```

Численно найдем и решение исходной краевой задачи

```

> exact_sol3 := dsolve( {ode3, bc3}, y(x), numeric,
output=listprocedure );

```

```

exact_sol3 := [ x = (proc (x) ... end proc ), y(x) = (proc (x) ... end proc ),
               d/dx y(x) = (proc (x) ... end proc ) ]
> y_exact := eval(y(x), exact_sol3);
               y_exact := proc (x) ... end proc


```

Сравним полученные решения

```

> pex3:=plot(y_exact(x), x=0..1, color=BLACK,
            legend=`краевая задача`);
psh3:=plot(y_shoot(x), x=0..1, color=BLACK,
            style=point, legend=`метод пристрелки`);
plots[display]({pex3, psh3});

```



И в этом случае мы имеем хорошее совпадение решений. Значение решения, полученное на правой границе методом пристрелки, есть

```

> evalf( y_shoot(1) );
-5.000000000

```

Метод конечных разностей

Для применения метода конечных разностей промежуток $[a, b]$, где требуется найти решение, узлами $x_0 = a, x_1 = a + h, \dots, x_n = b$, где $h = (b - a)/n$, разбивается на частичные промежутки. Дифференциальное уравнение, записанное для каждой точки x_i , заменяется конечно-разностным уравнением, и производные, входящие в краевые условия, — конечными разностями. В результате приходим к алгебраической задаче.

Снова рассмотрим для определенности краевую задачу (7.6). Заменяем производные неизвестной функции во внутренних точках с помощью центральных разностных производных по формулам

$$\left. \frac{dy}{dx} \right|_{x=x_k} \approx \frac{y_{k+1} - y_{k-1}}{2h}, \quad \left. \frac{d^2y}{dx^2} \right|_{x=x_k} \approx \frac{y_{k+1} - 2y_k + y_{k-1}}{h^2}, \quad k = 0, 1, \dots, n.$$

Отметим, что граничные условия дают $y_0 = r$, $y_n = s$, так что мы имеем $(n-1)$ уравнение относительно неизвестных y_1, y_2, \dots, y_{n-1} .

Для иллюстрации метода снова решим задачу 7.3.

```
> ode1 := diff( y(x), x$2 ) + 3 * diff( y(x), x ) +
7 * y(x) = 2*exp(-x) - 5;
bc1 := y(0)=2, y(1)=-5;
```

$$ode1 := \left(\frac{d^2}{dx^2} y(x) \right) + 3 \left(\frac{d}{dx} y(x) \right) + 7 y(x) = 2 e^{(-x)} - 5$$

$$bc1 := y(0) = 2, y(1) = -5$$

Определяем граничные точки,

```
> a := 0; b := 1;
```

$$a := 0$$

$$b := 1$$

Задаем количество точек и шаг

```
> N := 10; h := (b-a)/N;
```

$$h := \frac{1}{10}$$

Определяем узлы

```
> X := k -> a+k*h;
```

$$X := k \rightarrow a + k h$$

и конечно-разностные операторы

```
> Yp := k -> (y[k+1]-y[k-1])/2/h;
```

$$Yp := k \rightarrow \frac{1}{2} \frac{y_{k+1} - y_{k-1}}{h}$$

```
> Ypp := k -> (y[k+1]-2*y[k]+y[k-1])/h^2;
```

$$Ypp := k \rightarrow \frac{y_{k+1} - 2 y_k + y_{k-1}}{h^2}$$

Граничные условия обеспечивают нам два условия

```
> eq[0] := y[0] = rhs(bc1[1]);
```

$$eq_0 := y_0 = 2$$

```
> eq[N] := y[N] = rhs(bc1[2]);
      eq10 := y10 = -5
```

Формируем теперь уравнения для внутренних точек

```
> for k from 1 to N-1 do
      eq[k] := eval( odel,
                    {x=X(k), y(x)=y[k],
                     diff(y(x), x)=Yp(k),
                     diff(y(x), x$2)=Ypp(k)} );
end do;
```

$$eq_1 := 115 y_2 - 193 y_1 + 85 y_0 = 2 e^{\left(\frac{-1}{10}\right)} - 5$$

$$eq_2 := 115 y_3 - 193 y_2 + 85 y_1 = 2 e^{(-1/5)} - 5$$

$$eq_3 := 115 y_4 - 193 y_3 + 85 y_2 = 2 e^{\left(\frac{-3}{10}\right)} - 5$$

$$eq_4 := 115 y_5 - 193 y_4 + 85 y_3 = 2 e^{(-2/5)} - 5$$

$$eq_5 := 115 y_6 - 193 y_5 + 85 y_4 = 2 e^{(-1/2)} - 5$$

$$eq_6 := 115 y_7 - 193 y_6 + 85 y_5 = 2 e^{(-3/5)} - 5$$

$$eq_7 := 115 y_8 - 193 y_7 + 85 y_6 = 2 e^{\left(\frac{-7}{10}\right)} - 5$$

$$eq_8 := 115 y_9 - 193 y_8 + 85 y_7 = 2 e^{(-4/5)} - 5$$

$$eq_9 := 115 y_{10} - 193 y_9 + 85 y_8 = 2 e^{\left(\frac{-9}{10}\right)} - 5$$

Даже тогда, когда система линейна, лучше использовать команду `fsolve` для нахождения приближенного решения

```
> fd_sol1:=fsolve({seq(eq[k], k=0..N)}, {seq(y[k], k=0..N)});
fd_sol1 := {y3 = -8.221026622, y4 = -9.332266223, y1 = -2.726601601,
            y2 = -6.081951602, y7 = -8.492393882, y6 = -9.275651789, y5 = -9.617386766,
            y9 = -6.230467627, y8 = -7.431377798, y10 = -5., y0 = 2.}
```

Теперь переформатируем наше решение для последующего графического отображения и табличного представления

```
> fd_table1 := eval([seq([X(k), y[k]], k=0..N)], fd_sol1):
```

Представим решение в виде таблицы

```
> linalg[stackmatrix]( fd_table1 );
```

0	2.
$\frac{1}{10}$	-2.726601601
$\frac{1}{5}$	-6.081951602
$\frac{3}{10}$	-8.221026622
$\frac{2}{5}$	-9.332266223
$\frac{1}{2}$	-9.617386766
$\frac{3}{5}$	-9.275651789
$\frac{7}{10}$	-8.492393882
$\frac{4}{5}$	-7.431377798
$\frac{9}{10}$	-6.230467627
1	-5.

В силу простоты рассматриваемой задачи, можно найти точное решение

```
> infolevel[dsolve] := 3:
```

```
> exact_sol1 := combine(dsolve( { odel, bcl }, y(x) ));
```

```
> infolevel[dsolve] := 0:
```

Methods for second order ODEs:

--- Trying classification methods ---

trying a quadrature

trying high order exact linear fully integrable

trying differential order: 2; linear nonhomogeneous with symmetry [0,1]

trying a double symmetry of the form [xi=0, eta=F(x)]

Try solving first the homogeneous part of the ODE

-> Tackling the linear ODE "as given":

checking if the LODE has constant coefficients

<- constant coefficients successful

<- successful solving of the linear ODE "as given"

-> Determining now a particular solution to the non-homogeneous ODE

building a particular solution using variation of parameters

<- solving first the homogeneous part of the ODE successful


$$\begin{aligned}
 exact_soll := y(x) = & \left(-14 e^{\left(-\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{19} x}{2}\right) \cosh(1) \right. \\
 & + 14 e^{\left(-\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{19} x}{2}\right) \sinh(1) - 81 e^{\left(-\frac{3x}{2}\right)} \cosh\left(\frac{3}{2}\right) \sin\left(\frac{\sqrt{19} x}{2} - \frac{\sqrt{19}}{2}\right) \\
 & + 81 e^{\left(-\frac{3x}{2}\right)} \sinh\left(\frac{3}{2}\right) \sin\left(\frac{\sqrt{19} x}{2} - \frac{\sqrt{19}}{2}\right) - 150 e^{\left(-\frac{3x}{2}\right)} \sin\left(\frac{\sqrt{19} x}{2}\right) \\
 & + 14 e^{(-x)} \sin\left(\frac{\sqrt{19}}{2}\right) \cosh\left(\frac{3}{2}\right) - 14 e^{(-x)} \sin\left(\frac{\sqrt{19}}{2}\right) \sinh\left(\frac{3}{2}\right) \\
 & \left. - 25 \sin\left(\frac{\sqrt{19}}{2}\right) \cosh\left(\frac{3}{2}\right) + 25 \sin\left(\frac{\sqrt{19}}{2}\right) \sinh\left(\frac{3}{2}\right) \right) / \left(\right. \\
 & \left. 35 \sin\left(\frac{\sqrt{19}}{2}\right) \cosh\left(\frac{3}{2}\right) - 35 \sin\left(\frac{\sqrt{19}}{2}\right) \sinh\left(\frac{3}{2}\right) \right)
 \end{aligned}$$

Для сравнения решений отобразим их на одном графике

```

> pex:=plot(rhs(exact_soll),x=a..b,color=BLACK,
  legend=`точное решение`):
psh:=plot(fd_table1,x=a..b,color=BLACK,
  style=point,legend=`конечно-разностное решение`):
plots[display]({pex,psh});

```



Графики показывают хорошее согласие найденных решений.

Численное решение дифференциальных уравнений в частных производных

Система Maple может находить численные решения линейных систем дифференциальных уравнений в частных производных по прямоугольным областям.

Так же как и в случае обыкновенных дифференциальных уравнений, для численного решения уравнения (или системы уравнений) в частных производных служит универсальная функция `pdsolve/numeric`. Вызов этой функции осуществляется в следующем формате:

```
pdsolve (PDEsys, conditions, numeric, vars, options);
```

Параметры функции:

- `PDEsys` — одиночное уравнение или набор или список *зависящих от времени* линейных дифференциальных уравнений в частных производных от *двух* независимых переменных;
- `conditions` — набор или список начальных и граничных условий;
- `numeric` — ключевое слово, указывающее на необходимость получения численного решения;
- `vars` — (необязательный) зависимая переменная или набор или список зависимых переменных для `PDEsys`;
- `options` — (необязательные) уравнения ключевого слова формы `keyword = value`, где `keyword` — ключевое слово — одно из `'indepvars'`, `'time'`, `'range'`, `'spacestep'`, `'timestep'`, `'bcopts'`, или `'optimize'`; определяет опции для задачи и ее решения.

Функция

```
pdsolve(PDEsys, conditions, numeric, vars, options);
```

возвращает модуль, который может использоваться для определения численного решения зависящих от времени (то есть гиперболических или параболических) уравнений в частных производных или систем таких уравнений по простым прямоугольным областям. Функция `pdsolve/numeric` — стандартная программа, использующая обычные методы конечных разностей для получения численного решения. Уравнение PDE или система уравнений `PDEsys` должна быть определена в `diff` — системе обозначений. Производные в начальных/граничных условиях должны быть определены в индексированной `D` — системе обозначений (например, для $u(x, y)$, `D[1](u)(0, y)` описывает производную по нормали на линии $x = 0$).

Параметр `vars` определяет зависимую переменную (или переменные) задачи. Для решающих программ (`solvers`), которые могут вычислять решения систем уравнений в частных производных, это указывает порядок для зависимых переменных на выходе.

Необязательное уравнение `'indepvars'=list` определяет независимые переменные задачи, и указывает порядок независимых переменных для возвращаемых процедур, которые вычисляют решение в данных значениях независимых переменных.

Решающая программа имеет два режима работы. Первый использует заданный по умолчанию метод, который является централизованной неявной схемой, и позволяет получать решения линейного уравнения или линейной системы уравнений высокого порядка. Уравнения системы должны быть достаточно близки к стандартной форме для метода нахождения численного решения.

Второй режим работы строго образовательный, позволяет указывать специальные методы решения уравнения. Этот режим ограничен решением одного единственного линейного уравнения в частных производных.

Оба режима допускают классические граничные условия 1-го, 2-го, 3-го и 4-го рода, то есть граничные условия Дирихле, Неймана, Робина, или периодические граничные условия.

Необязательные уравнения (в дополнение к 'indepvars') для заданного по умолчанию метода следующие:

- 'time' = name;
- 'range' = l .. r;
- 'spacestep' = numeric;
- 'timestep' = numeric;
- 'bcpts' = integer;
- 'optimize' = boolean/symbol (булева переменная/символ).

Опция time определяет имя переменной времени для задачи. Параметр range определяет пространственную область (домен) задачи. Значения обеих опций определены автоматически, если граничные условия назначены для обеих конечных точек домена. Эти опции необходимы только для аperiodических задач первого порядка, где только одна из конечных точек появляется в условии.

Опция spacestep определяет интервалы для пространственных точек на дискретной сетке, на которой вычисляется решение; значения по умолчанию 1/20-ая из пространственного диапазона задачи range, где l – левосторонняя граница, и r – правосторонняя граница. Интервал должен быть малым, чтобы достаточное число точек находилось в пространственной области для данного метода, граничных условий, и пространственной интерполяции. Если данное значение spacestep не вписывается во входной домен с целым числом времен, выбирается самый близкий размер шага, который может быть использован.

Опция timestep задает размер шага по времени, используемого при интегрировании. Значение по умолчанию соответствует значению пространственного шага spacestep.

Параметр bcpts определяет число точек, использованных в дискретизации граничных условий задачи. Это значение по умолчанию на единицу больше чем порядок производных по пространственным переменным — первый порядок точности аппроксимации граничного условия.

Параметр `optimize` определяет уровень оптимизации. Три типа оптимизации доступны: `true`, `'partial'`, и `false`. Заданная по умолчанию `'partial'` оптимизация, как отмечают разработчики, дает довольно хорошие результаты для малых времен. Установка `false` блокирует оптимизацию в целом (не рекомендуемый параметр). Установка `true` заставляет `pdsolve` вызывать, `codegen[optimize]` при построении численного метода. Применение полностью оптимизированного метода приводит к дополнительным затратам времени, поэтому и рекомендуется `'partial'` — настройка по умолчанию.

На этом мы заканчиваем краткое описание функции `pdsolve/numeric`. Более полную информацию можно получить по справке `?pdsolve, numeric`.

Рассмотрим применение этой функции для численного решения задач математической физики.

Задача 7.6. Решить следующую смешанную задачу для параболического уравнения:

$$\frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2} - 2 \frac{\partial u}{\partial x} + u + e^x \sin(x) - t, \quad 0 < x < \pi, \quad t > 0,$$

$$u|_{x=0} = 1 + t, \quad u|_{x=\pi} = 1 + t, \quad u|_{t=0} = 1 + e^x \sin(2x).$$

Решение. Определяем уравнение и дополнительные условия задачи:

```
> restart:
> PDE:=
diff(u(x,t),t)-diff(u(x,t),x,x)+2*diff(u(x,t),x)-u(x,t)=
exp(x)*sin(x)-t;
IBC:={u(0,t)=1+t,u(Pi,t)=1+t,u(x,0)=1+exp(x)*sin(2*x)};
PDE:=\left(\frac{\partial}{\partial t}u(x,t)\right)-\left(\frac{\partial^2}{\partial x^2}u(x,t)\right)+2\left(\frac{\partial}{\partial x}u(x,t)\right)-u(x,t)=e^x\sin(x)-t
IBC:={u(0,t)=1+t,u(\pi,t)=1+t,u(x,0)=1+e^x\sin(2x)}
```

Точное решение этой задачи дается функцией

$$v(x,t) = 1 + t + (1 - e^{-t})e^x \sin(x) + e^{x-4t} \sin(2x),$$

то есть

```
> v:=(x,t)->t+1+(1-exp(-t))*exp(x)*sin(x)+
exp(x-4*t)*sin(2*x);
v:=(x,t)->t+1+(1-e^(-t))*e^x*sin(x)+e^(x-4*t)*sin(2*x)
```

Действительно, проверим это

```
> simplify(subs(u(x,t)=v(x,t),PDE));
e^x*sin(x)-t=e^x*sin(x)-t
```

```
> v(0, t); v(Pi, t); v(x, 0);
1 + t
1 + t
1 + e^x sin(2 x)
```

Мы рекомендуем читателю в качестве упражнения получить это точное решение.

Решим задачу численно с помощью функции `pdsolve` с установленной опцией `numeric`:

```
> pds := pdsolve(PDE, IBC, numeric);
pds := module() export plot, plot3d, animate, value, settings; ... end module
```

Модуль, возвращаемый функцией `pdsolve/numeric`, предоставляет несколько способов, которые могут использоваться для вычисления или визуализации решения исходного уравнения или системы: `plot`, `plot3d`, `animate`, и `value`. Кроме того, `settings` делает возможным настройку параметров процесса решения.

Если результат, возвращаемый функцией `pdsolve/numeric`, присвоен переменной `pde`, вызывающие последовательности для использования различных способов представления решения имеют форму `pde:-method (arguments)`, где `method` — один из `plot`, `plot3d`, `animate`, `value`, или `settings`.

Для наглядного сравнения точного решения с численным решением построим графики:

```
> p1:=pds:-plot(t=0.5, title=`Профиль температуры при
t=0.5`, style=point, symbol=cross, legend=`Численное решение,
t=0.5`, color=black):
> q1:=plot(v(x, 0.5), x=0..Pi, color=black, linestyle=3,
legend=`Точное решение, t=0.5`):
> plots[display]({p1, q1});
```



Посмотрим еще графики, например, при $t = 10$:

```
> p3:=pds:-plot(t=10, title=`Профиль температуры при
t=10`, style=point,
```

```

symbol=circle,color=black,legend=`Численное решение,
t=10`):
q3:=plot(v(x,10),x=0..Pi,color=black,legend=`Точное
решение, t=10`):
plots[display]({p3,q3});

```




Можно посмотреть пространственную поверхность температур:

```

> pds:-plot3d(t=0..10,x=0..Pi,axes=boxed,
orientation=[-145,45],
color=[3,10,u]);

```




Определим, например, при каких значениях x в момент времени $t = 5$ температура будет равна $u(x,5) = 10$. Для этого построим профиль температур, чтобы отделить корни

```

> pp1:=pds:-plot(t=5,title=`Профиль температуры при t=5`,
color=black,style=point,symbol=cross):
pp2:=plot(10,x=0..Pi,color=black):
plots[display]({pp1,pp2});

```



Из графика видно, что это будут диапазоны $1 < x < 2$ и $2,5 < x < \pi$. Теперь, чтобы получить численные значения, мы создадим специальную процедуру, используя процедуру `value`, имеющуюся в модуле решения

```
> pds:-value (t=5, output=listprocedure);
uval := rhs (op (3, %));
[x = (proc(x) ... end proc), t = 5., u(x, t) = (proc(x) ... end proc)]
uval := proc(x) ... end proc
```

И, наконец, найдем нужные нам значения координаты x :

```
> fsolve (uval (x) = 10, x = 1..2.); fsolve (uval (x) = 10, x = 2.5..Pi);
1.400565166
2.928179056
```

Итак, температура будет равна 10 при $t = 5$ в двух точках: $x \approx 1,4$ и $x \approx 2,9$.

Можно также определить температуру в любой точке, например, в точке с координатами $x = 2,5$ и $t = 7$:

```
> U := pds:-value (); U (2.5, 7.); v (2.5, 7.);
U := proc() ... end proc
[x = 2.5, t = 7., u(x, t) = 15.3872910100265976]
15.28423485
```

Итак, $u(2,5;7) \approx 15,4$. Точное значение — $v(2,5;7) \approx 15,3$.

Рассмотрим еще пример решения гиперболического уравнения.

Задача 7.7. Исследовать малые поперечные колебания струны, закрепленной в своих концах при заданном начальном смещении и нулевой начальной скорости.

Решение. Задача, как известно, сводится к интегрированию одномерного волнового уравнения при дополнительных условиях:

$$\frac{\partial^2 u}{\partial t^2} = \frac{1}{c^2} \frac{\partial^2 u}{\partial x^2}, \quad 0 < x < l, \quad t > 0,$$

$$u|_{x=0} = 0, \quad u|_{x=l} = 0, \quad u|_{t=0} = f(x), \quad \left. \frac{\partial u}{\partial x} \right|_{x=0} = 0.$$

Воспользуемся функцией `Laplacian`, имеющейся в Maple в пакете `VectorCalculus`. Эта функция полезна для записи дифференциальных уравнений в частных производных 2-го порядка. Сначала загружаем этот пакет:

```
> with(VectorCalculus):
```

Определяем уравнение и дополнительные условия задачи (будем считать для определенности $c = \sqrt{10}$, $l = 1$):

```
> Cord_Eqn := {diff(u(x,t), t, t) =
  Laplacian(u(x,t), 'cartesian'[x])/10};
  Cord_Eqn := {∂²u(x,t)/∂t² = 1/10 (∂²u(x,t)/∂x²)}
> BC1 := {u(0,t) = 0, u(1,t) = 0, D[2](u)(x,0) = 0};
  BC1 := {u(0,t) = 0, u(1,t) = 0, D₂(u)(x,0) = 0}
```

Определим теперь начальную функцию $f(x)$. Рассмотрим случай, когда струна в начальный момент времени $t = 0$ оттянута в центре и затем отпущена без начальной скорости. Будем иметь:


```
> BC2 := {u(x,0) =
  piecewise(x >= 0 and x <= 1/2, x, x > 1/2 and x <= 1, 1-x)};
  BC2 := {u(x,0) = { x -x ≤ 0 and x - 1/2 ≤ 0
                    -x + 1 -x < -1/2 and x - 1 ≤ 0 }}
```

Решаем начально-краевую задачу численно с использованием функции `pdsolve` и установленной опцией `numeric`. Мы можем также установить точность решения, определяя шаг по времени и по пространственной переменной по прямоугольнику «расстояние-время»:

```
> sol := pdsolve(Cord_Eqn, BC1 union BC2, numeric,
  spacestep = 1/200, timestep = 1/100):
```

Посмотрим профиль струны в движении:

```
> sol := animate(u(x,t), t = 0..2*Pi, frames = 30,
  labels = ["x", "u(x,t)"], labelfont = [TIMES, ROMAN, 14],
  scaling = constrained);
```

На рисунке, конечно, отображен только один фрагмент и, чтобы увидеть колебания струны в движении, надо запустить анимацию, нажав соответствующую кнопку на панели инструментов.

Рассмотрим теперь решение задачи при другой начальной функции. Пусть, например, струна имеет начальную форму в виде тригонометрической функции, скажем,

$$f(x) = \frac{1}{2} \sin(4\pi x):$$

```
> BC2 := {u(x,0)=sin(4*Pi*x)/2};
```

$$BC2 := \{u(x,0) = \frac{1}{2} \sin(4\pi x)\}$$


Численно решаем задачу

```
> sol:=pdsolve(Cord_Eqn,BC1 union BC2,numeric,
  spacestep=1/200,timestep=1/100);
```

```
sol := module() export plot, plot3d, animate, value, settings; ... end module
```

и наблюдаем движение струны


```
> sol:-animate( u(x,t), t=0..Pi/2, frames=30,
  labels=["x", "u(x,t)"],
  labelfont=[TIMES,ROMAN,14],
  scaling=constrained);
```



Можно также отображать несколько функций решения на одном графике, например,

```
> sol:-animate( [[u(x,t) + sin(120*x)/10, color=red],
```

```
[-u(x,t)/2 + sin(120*x)/20, color=blue]], t=0..Pi/2,
frames=30, labels=["x", "u(x,t)"], labelfont=[TIMES, ROMAN, 14],
scaling=constrained);
```



Все рассмотренные задачи могут быть решены в более ранней версии Maple 8. Скажем несколько слов о возможностях новых версий, начиная с Maple 9 и старше, в разрезе решения дифференциальных уравнений. Как отмечают разработчики, в новых версиях улучшены решающие программы, как для обыкновенных дифференциальных уравнений, так и для уравнений в частных производных, как для точного, так и для численного решения. Это позволяет обрабатывать больше классов задач. Появились новые методы решения для нелинейных краевых задач классической математической физики, для систем второго порядка, допускающие гипергеометрические решения, и многое другое.

В качестве демонстрации численного решения нелинейного уравнения рассмотрим одномерный вариант уравнения Бюргерса

$$\frac{\partial u}{\partial t} + u \frac{\partial u}{\partial x} - \frac{1}{\text{Re}} \frac{\partial^2 u}{\partial x^2}, \quad (7.8)$$

где Re — число Рейнольдса.

Это уравнение представляет собой хорошую модель уравнений Навье – Стокса, так как оно в простейшей форме описывает баланс между нелинейным конвективным процессом (член $u \partial u / \partial x$) и диссипативным процессом (член $(1/\text{Re}) \partial^2 u / \partial x^2$). Несмотря на то, что уравнение Бюргерса нелинейно, для многих сочетаний начальных и граничных условий оно допускает точное решение. В силу этой причины уравнение Бюргерса часто используется для выяснения точности, обеспечиваемой вычислительными алгоритмами.

Рассмотрим задачу о распространяющейся ударной волне, которая первоначально представляет собой разрыв, расположенный в плоскости $x = 0$. В дальнейшем эта волна распространяется вправо, однако ее профиль, бывший вначале разрывным, сглаживается за счет воздействия вязкого диссипативного процесса $(1/\text{Re}) \partial^2 u / \partial x^2$.

Будем искать решение уравнения (7.8) в области $-1 \leq x \leq 1$ при $t \geq 0$. Начальные и граничные условия примем следующими

$$u(x, 0) = u_0(x) = \begin{cases} 1, & -1 \leq x \leq 0, \\ 0, & 0 < x \leq 1, \end{cases}$$

$$u(-1, t) = 1, \quad u(1, t) = 0.$$

Определяем уравнение в Maple:

```
> eq:=diff(u(x,t),t)+u(x,t)*diff(u(x,t),x)-
1/R*diff(u(x,t),x$2)=0;
```

$$eq := \left(\frac{\partial}{\partial t} u(x, t) \right) + u(x, t) \left(\frac{\partial}{\partial x} u(x, t) \right) - \frac{\partial^2 u(x, t)}{R} = 0$$

Определяем начальные и граничные условия:

```
> ibc:={u(x,0)=u0(x), u(-1,t)=1, u(1,t)=0};
      ibc:={u(x,0)=u0(x), u(-1,t)=1, u(1,t)=0}
```

Определяем начальный профиль волны:

```
> u0:=x->1-Heaviside(x);
      u0:=x -> 1 - Heaviside(x)
```

Для последующего отображения решения на графиках подготовим график этого начального профиля:

```
> ph:=plot(u0(x), x=-1..1, color=blue,
           linestyle=3, legend='t=0', labels=['x', 'u']):
```

Подготовим также надписи на осях графиков:

```
> pt:=textplot([0.1, 1, 'u'], [1, 0.05, 'x']):
```


Задаем значение числа Рейнольдса и решаем задачу:

```
> R:=10; pds := pdsolve(eq, ibc, numeric);
      R:=10
```

```
pds := module() export plot, plot3d, animate, value, settings; ... end module
```

Отобразим решение на графике в момент времени $t = 0,47$

```
> p_10:=pds:-plot(t=0.47, numpoints=50, color=black,
                 thickness=2, legend='Re=10, t=0.47',
                 scaling=constrained, labels=['x', 'u'],
                 axes=frame):
plots[display]({p_10, ph, pt});
```



Найдем теперь решение для другого значения числа Рейнольдса

```
> R:=50;pds_50 := pdsolve(eq,ibc,numeric);
```


```
      R := 50
```

```
      pds_50 := module() export plot, plot3d, animate, value, settings; ... end module
```

Отообразим результат на графике в момент времени $t = 0,36$


```
> p_50:=pds_50:-plot(t=0.36,numpoints=50,color=brown,
      thickness=2,legend=`Re=50, t=0.36`,
      scaling=constrained,labels=[``,``],
      axes=frame):
```

```
> plots[display]({p_50,ph,pt});
```



Решение в области скачка неудовлетворительно. Попытаемся улучшить его за счет дробления сетки, на которой ищется решение


```
pds_50_ := pdsolve(eq, ibc, numeric, spacestep=1/40);  
pds_50_ := module() export plot, plot3d, animate, value, settings; ... end module  
> p_50_ := pds_50_:-plot(t=0.36, numpoints=50, color=brown,  
    thickness=2, legend=`Re=50, t=0.47`,  
    scaling=constrained, labels=[``, ``],  
    axes=frame):  
> plots[display]({p_50_, ph, pt});
```



Теперь все в порядке!


При уточнении решения мы заново выполнили расчет с другим шагом по координате x . Улучшения решения можно добиться, используя уже полученное решение `pds_50`, установив опцию `settings`, в которой зададим шаг по координате x :

```
> pds_50:-settings(spacestep=1/40);  
> p_50n := pds_50:-plot(t=0.36, numpoints=50, color=black,  
    style=point, thickness=2,  
    legend=`Re=50, t=0.36, spacestep=1/40`,  
    scaling=constrained, labels=[``, ``],  
    axes=frame):  
plots[display]({p_50n, ph, pt, p_50_});
```



Еще лучший результат мы получим, измельчив сетку, и по координате, и по времени:

```
> pds_50:-settings(timestep=1/40,spacestep=1/40);
> p_50nn:=pds_50:-plot(t=0.36,numpoints=50,color=black,
    style=point,symbol=cross,thickness=2,
    legend=`Re=50, t=0.36,timestep=1/40,spacestep=1/40`,
    scaling=constrained,labels=[``,``],axes=frame):
> plots[display]({p_50n,ph,pt,p_50_,p_50nn});
```



Упражнения

1. Применяя метод Эйлера, численно решить данные дифференциальные уравнения с данными начальными условиями на отрезке $[a, b]$ с шагом $h = 0,1$:

1.1. $y' = \frac{1}{2}xy, y(0) = 1, a = 0, b = 1$;

1.2. $y' = x^2 + y^2, y(0) = 0, a = 0, b = 1$;

1.3. $y' = 1 + xy^2, y(0) = 0, a = 0, b = 1$;

1.4. $y' = \frac{y}{x+1} - y^2, y(0) = 1, a = 0, b = 1$;

1.5. $y' = \alpha y^2 + \frac{\beta}{x^2}, y(1) = 1, a = 1, b = 2$,

а) $\alpha = -1, \beta = 0,05 + 0,05k, k = 0,1,2,\dots,7$

б) $\alpha = -0,5, \beta = 0,1 + 0,1k, k = 0,1,2,3$.

2. Применяя один из улучшенных методов Эйлера, найти значение решения дифференциального уравнения с данным начальным условием и данным шагом в указанной точке:

2.1. $y' = x + y^2, y(0) = 0, h = 0,03$, найти $y(0,3)$;

2.2. $y' = 1 + x - y^2, y(0) = 1, h = 0,02$, найти $y(0,1)$;

2.3. $y' = x^2 - y^2, y(0) = 0, h = 0,1$, найти $y(1)$;

2.4. $y' = xy - 0,1y^2, y(0) = 0, h = 0,02$, найти $y(0,2)$;

2.5. $y' = x^3 + y^2, y(0) = 0,5, h = 0,1$, найти $y(0,5)$;

2.6. $y' = x + \sqrt{y}, y(0,5) = 0,7240, h = 0,1$, найти $y(1,5)$;

2.7. $y' = 2x + \cos y, y(0) = 0, h = 0,02$, найти $y(0,1)$;

2.8. $y' = e^x - y^2, y(0) = 0, h = 0,04$, найти $y(0,4)$;

2.9. $y' = x \ln y - y \ln x, y(1) = 1, h = 0,1$, найти $y(1,6)$;

2.10. $y' = 0,1 \left[\sqrt[3]{y} + \ln(x+y) - 1 \right], y(-1) = 2, h = 0,2$, найти $y(1)$.

3. Методом Рунге-Кутты с шагом $h = 0,2$ найти решения данных уравнений на указанном отрезке $[a, b]$:

3.1. $y' = -x + y, y(0) = 1,5, a = 0, b = 1$;

3.2. $y' = \frac{y}{x} - y^2, y(1) = 1, a = 1, b = 2$;

4. Методом Рунге-Кутты с шагом $h = 0,1$ найти решение данного уравнения на отрезке $[0; 0,3]$ при заданных значениях параметров и заданном начальном условии:

$$y' = \frac{\cos bx}{a + y^2}, y(0) = 1,5, a = 1,0 + 0,4n, b = 1,0 + 0,8k, n = 0,1,2,3,4,5, k = 0,1,2.$$

5. Методом Рунге-Кутты с шагом $h = 0,1$ найти решение данного уравнения на отрезке $[0; 0,3]$ при заданных значениях параметров и заданном начальном условии:

$$y' = e^{-ax} (y^2 + b), y(0) = 0, a = 1,0 + 0,4n, b = 1,0 + 0,4k, \\ n = 0,1, \dots, 5, k = 0,1, \dots, 5.$$

6. Методом конечных разностей найти решение краевой задачи:

$$x^2 y'' + xy' = 1, y(1) = 0, y(1,4) = \frac{1}{2} \ln^2(1,4) = 0,0566.$$

7. Методом конечных разностей найти решение краевой задачи:

$$y'' - 2xy' - 2y = -4x, y(0) - y'(0) = 0, y(1) = 1 + e = 3,718.$$

8. Методом конечных разностей найти решение краевой задачи:

$$y'' - 2xy' - 2y = -4x, y(0) - y'(0) = 0, 2y(1) - y'(1) = 1.$$

9. Методом конечных разностей найти решение краевой задачи с точностью до 0,001:

$$y'' + 2xy' + 2y = 4x, y(0) = 1, y(0,5) = e^{-0,25} + 0,5 = 1,279.$$

10. Найти численное решение уравнения $\frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2}$, удовлетворяющее условиям $u(x,0) = f(x), u(0,t) = \varphi(t), u(1,t) = \psi(t)$, для значений $0 \leq t \leq T$, взяв по аргументу x шаг $h = 0,1$. Задачу решить при следующих данных:

$$a) f(x) = (ax^2 + b) \sin \pi x, \varphi(t) = \psi(t) = 0,$$

$$T = 0,02, a = 1,1; 1,3; 1,5, b = 1,1 + 0,1n, n = 0,1,2,3,4;$$

$$б) f(x) = e^{-bx} \sin ax, \varphi(t) = 0, \psi(t) = e^{-b} \sin a,$$

$$T = 0,02, a = \frac{\pi}{12}; \frac{\pi}{4}; \frac{\pi}{3}, b = 0,1n, n = 0,1,2,3,4,5.$$

11. Найти численное решение уравнения $\frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2} + 2x + t$, удовлетворяющее условиям $u(x,0) = f(x)$, $u(0,t) = \varphi(t)$, $u(1,t) = \psi(t)$, для значений $0 \leq t \leq T$, взяв по аргументу x шаг $h = 0,1$. Задачу решить при следующих данных:

$$f(x) = (ax^2 + b) \sin \pi x, \varphi(t) = \psi(t) = 0,$$

$$T = 0,02, a = 1,1; 1,3; 1,5, b = 1,1 + 0,1n, n = 0,1,2,3,4.$$

12. Найти численное решение уравнения $\frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2} + 3t \sin x$, удовлетворяющее условиям $u(x,0) = f(x)$, $u(0,t) = \varphi(t)$, $u(1,t) = \psi(t)$, для значений $0 \leq t \leq T$, взяв по аргументу x шаг $h = 0,1$. Задачу решить при следующих данных:

$$f(x) = e^{-bx} \sin ax, \varphi(t) = 0, \psi(t) = e^{-b} \sin a,$$

$$T = 0,02, a = \frac{\pi}{12}; \frac{\pi}{4}; \frac{\pi}{3}, b = 0,1n, n = 0,1,2,3,4,5.$$

13. Найти численное решение уравнения $\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}$, удовлетворяющее условиям $u(x,0) = 0, 2x(1-x) \sin \pi x$, $u_t(x,0) = 0$, $u(0,t) = 0$, $u(1,t) = 0$.

14. Найти численное решение уравнения $\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}$, удовлетворяющее условиям $u(x,0) = x(\pi - x)$, $u_t(x,0) = 0$, $u(0,t) = 0$, $u(\pi,t) = 0$.

15. Найти численное решение уравнения $\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}$, удовлетворяющее условиям $u(x,0) = f(x)$, $u_t(x,0) = 0$, $u(0,t) = 0$, $u(1,t) = 0$. Задачу решить при следующих данных: $f(x) = (ax^2 + 1,1) \sin \pi x$, $a = 1,1 + 0,1n$, $n = 0,1,2,3,4$.

16. Найти численное решение уравнения $\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + x + t$ с шагом $h = l = 0,1$, удовлетворяющее условиям $u(x,0) = (1,5x^2 + 1,2) \sin \pi x$, $u_t(x,0) = 0,1x$, $u(0,t) = 0$, $u(1,t) = 0$.

17. Найти численное решение уравнения $\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} + x^2 - t^2$ с шагом $h = l = 0,1$, удовлетворяющее условиям $u(x, 0) = (1,5x^2 + 0,9)e^{-x}$, $u_t(x, 0) = 0$, $u(0, t) = 0$, $u(1, t) = 2,4e^{-1}$.

Методы взвешенных невязок

Большинство инженерных задач, которые описываются сложными дифференциальными уравнениями, могут быть решены лишь приближенными методами, например, численными. К наиболее хорошо изученным численным методам относится метод конечных разностей, с которым мы уже кратко познакомимся.

Метод конечных разностей или метод сеток был одним из первых приближенных методов численного решения граничных задач. В методе сеток разрешающие уравнения аппроксимируются с помощью локальных разложений неизвестных функций в ряды, как правило, в усеченные ряды Тейлора. Оказывается, что такой прием можно интерпретировать как частный случай общего метода взвешенных невязок.

Настоящая глава ставит своей целью описание различных приближенных методов решения граничных задач и способов реализации этих методов в Maple. Здесь рассматриваются методы, применение которых восходит ко времени, когда еще не было вычислительной техники; к ним относятся методы Галеркина, коллокаций, наименьших квадратов, Рунге и другие.

Непрерывный прогресс в вычислительной технике на некоторое время отвлек исследователей от развития аналитических методов. На протяжении десятилетий использования вычислительной техники численные методы фактически вытеснили из практики аналитические методы решения технических задач. Системы аналитических вычислений могут в корне изменить отношение к “забытым” аналитическим методам. В этой главе при изложении основ методов мы будем следовать работе [8].

Аппроксимация функций

Ключ к проблеме численного решения дифференциальных уравнений лежит в возможности получения методов аппроксимации функций. Поэтому рассмотрение приближенных методов решения граничных задач для дифференциальных уравнений мы начнем с описания способов аппроксимации функций.

В задачах, описываемых дифференциальными уравнениями, как правило, требуется найти решение, удовлетворяющее определенным заданным граничным условиям. Поэтому рассмотрим сначала методы построения аппроксимаций, которые на границе Γ некоторой области Ω принимали бы те же значения, что и некоторая заданная функция φ .

Допустим, нам удалось найти некоторую функцию ψ , принимающую на границе Γ одинаковые значения с функцией φ , то есть $\varphi|_{\Gamma} = \psi|_{\Gamma}$. Пусть, далее, мы выбрали систему линейно независимых базисных функций $\{N_m; m = 1, 2, \dots\}$, таких, что $N_m|_{\Gamma} = 0$ для всех m (такие функции иногда называют *координатными*). Тогда на множестве Ω можно предложить аппроксимацию для функции φ

$$\varphi \approx \tilde{\varphi} = \psi + \sum_{m=1}^M a_m N_m, \quad (8.1)$$

где a_m ($m = 1, 2, \dots, M$) — некоторые числовые параметры, вычисляемые таким образом, чтобы получить «хорошее приближение». Очевидно, аппроксимация (8.1) обладает свойством $\tilde{\varphi}|_{\Gamma} = \varphi|_{\Gamma}$ для любых значений параметров a_m . Ясно, что система базисных функций должна быть выбрана таким образом, чтобы гарантировать улучшение аппроксимации при возрастании числа M используемых базисных функций. Таким образом, система базисных функций должна обладать следующим свойством: правая часть равенства (8.1) при $M \rightarrow \infty$ может сколь угодно точно представлять произвольную функцию φ , удовлетворяющую условию $\varphi|_{\Gamma} = \psi|_{\Gamma}$. Это свойство системы базисных функций называется условием *полноты*.

Рассмотрим общий метод определения параметров a_m в представлении (8.1), а именно, рассмотрим аппроксимацию с помощью взвешенных невязок. Введем погрешность аппроксимации R_{Ω} или *невязку*, которую определим по правилу

$$R_{\Omega} = \varphi - \tilde{\varphi}. \quad (8.2)$$

Невязка R_{Ω} — есть функция точки области Ω . Чтобы уменьшить невязку на всей области Ω , потребуем равенства нулю соответствующего числа интегралов от погрешности, взятых с различными весами, то есть

$$\int_{\Omega} W_m R_{\Omega} d\Omega \equiv \int_{\Omega} W_m (\varphi - \tilde{\varphi}) d\Omega = 0; \quad m = 1, 2, \dots, M, \quad (8.3)$$

где $\{W_m; m = 1, 2, \dots\}$ — множество линейно независимых *весовых функций*. Тогда общее условие сходимости $\tilde{\varphi} \rightarrow \varphi$ при $M \rightarrow \infty$ можно записать в форме условия выполнения равенств (8.3) для всех m при $M \rightarrow \infty$. Это условие, в свою очередь, будет означать, что $R_{\Omega} \rightarrow 0$ во всех точках области Ω .

Подставляя в (8.3) выражение (8.1) для $\tilde{\varphi}$, видим, что *система уравнений метода взвешенных невязок* (8.3) сводится к системе линейных алгебраических уравнений для неизвестных коэффициентов a_m , которую можно записать в виде

$$\mathbf{Ka} = \mathbf{f}, \quad (8.4)$$

где

$$\mathbf{a}^T = (a_1, a_2, a_3, \dots, a_M),$$

$$K_{mn} = \int_{\Omega} W_m N_n d\Omega, \quad 1 \leq m, n \leq M, \quad (8.5)$$

$$f_m = \int_{\Omega} W_m (\varphi - \psi) d\Omega, \quad 1 \leq m \leq M. \quad (8.6)$$

Таким образом, если известна аппроксимируемая функция φ , определена функция ψ и выбраны подходящие системы базисных и весовых функций, то, решая уравнение (8.4), можно получить коэффициенты в аппроксимации (8.1).

На практике могут быть использованы различные системы весовых функций, что приводит к разным методам аппроксимации посредством метода взвешенных невязок. Рассмотрим некоторые из наиболее употребительных системы весовых функций.

Поточечная коллокация

Здесь элементы W_m системы весовых функций заданы формулой

$$W_m = \delta(x - x_m),$$

где $\delta(x - x_m)$ — дельта-функция Дирака, которая обладает, по определению, следующими свойствами

$$\delta(x - x_m) = \begin{cases} 0, & x \neq x_m, \\ \infty, & x = x_m; \end{cases} \quad \int_{-\infty}^{+\infty} f(x) \delta(x - x_m) dx = f(x_m).$$

Согласно (8.3), выбор таких весовых функций эквивалентен тому, что невязка R_Ω полагается равной нулю (то есть $\varphi = \tilde{\varphi}$) в ряде заданных точек x_m . Тогда матрица \mathbf{K} и вектор \mathbf{f} в (8.4) имеют элементы

$$K_{mn} = N_n|_{x=x_m}, \quad f_m = (\varphi - \psi)|_{x=x_m}. \quad (8.7)$$

Эти формулы естественно являются основой метода интерполяции — при таком подходе параметры a_m выбираются из условия, что аппроксимация $\tilde{\varphi}$ должна совпадать с функцией φ в M различных произвольно выбранных точках области Ω .

Коллокация по подобластям

Если весовые функции выбираются по правилу

$$W_m = \begin{cases} 1, & x_m < x < x_{m+1}, \\ 0, & x < x_m, \quad x > x_{m+1}, \end{cases}$$

то уравнения метода взвешенных невязок (8.3) равносильны требованию равенства нулю интеграла от погрешности по каждой из M подобластей основной области. Теперь, для завершения аппроксимации, остается решить систему (8.4), где

$$K_{mn} = \int_{x_m}^{x_{m+1}} N_n dx, \quad f_m = \int_{x_m}^{x_{m+1}} (\varphi - \psi) dx.$$

Метод Галеркина

Это один из наиболее популярных методов взвешенных невязок. Здесь, вместо привлечения новой системы функций, в качестве весовых множителей выбираются сами базисные функции, т.е. $W_m = N_m$. В таком случае в системе (8.4) матрица \mathbf{K} и вектор \mathbf{f} имеют элементы

$$K_{mn} = \int_{\Omega} N_m N_n d\Omega, \quad f_m = \int_{\Omega} N_m (\varphi - \psi) d\Omega. \quad (8.8)$$

Отметим, что симметричность матрицы \mathbf{K} обеспечивает методу вычислительные преимущества. Отметим также, что, используя для аппроксимации функции φ на отрезке $0 < x < L$ метод Галеркина с системой базисных функций

$$\left\{ N_m = \sin\left(\frac{m\pi x}{L}\right); m = 1, 2, \dots \right\},$$

в силу (8.8) приходим к коэффициентам

$$K_{mn} = \int_0^L \sin\left(\frac{m\pi x}{L}\right) \sin\left(\frac{n\pi x}{L}\right) dx = \begin{cases} \frac{L}{2}, & n = m, \\ 0, & n \neq m; \end{cases}$$

$$f_m = \int_0^L \sin\left(\frac{m\pi x}{L}\right) (\varphi - \psi) dx.$$

Таким образом, матрица системы (8.4) имеет диагональную форму, что позволяет сразу написать решение

$$a_m = \frac{2}{L} \int_0^L \sin\left(\frac{m\pi x}{L}\right) (\varphi - \psi) dx, \quad m = 1, 2, 3, \dots, M.$$

Отсюда видим, что представление функции посредством усеченных синус-рядов Фурье можно рассматривать как аппроксимацию по Галеркину с помощью взвешенных невязок. Особая простота уравнений, получаемых в этом случае, является следствием свойства *ортogonalности* используемых базисных функций, которое выражается равенством

$$\int_{\Omega} N_m N_n d\Omega = 0, \quad n \neq m.$$

Эта характеристика элементов системы базисных функций является чрезвычайно полезной.

Рассмотрим теперь на примере аппроксимацию заданной функции методом взвешенных невязок.

Задача 8.1. Используя подходящую систему базисных функций в виде многочленов, аппроксимировать функцию $\varphi = 1 + \sin\left(\frac{\pi x}{2}\right)$ на отрезке $0 \leq x \leq 1$. Применить метод поточечной коллокации и метод Галеркина. Численно исследовать сходимость аппроксимации к заданной функции.

Решение. Определим аппроксимируемую функцию в Maple

```
> phi := x -> 1 + sin(Pi*x/2);
```

$$\phi := x \rightarrow 1 + \sin\left(\frac{1}{2} \pi x\right)$$

Теперь выберем функцию ψ , удовлетворяющую граничным условиям. В качестве такой функции возьмем линейную функцию и тут же проверим справедливость граничных условий

```
> psi:=x->x+1;psi(0);psi(1);
      psi := x → x + 1
      1
      2
```

Выберем систему базисных функций и определим аппроксимирующую функцию; в качестве базисных функций возьмем систему полиномов, удовлетворяющих однородным граничным условиям на концах отрезка

```
> N:=(x,m)->x^m*(1-x);
Phi:=proc(x,M)option operator,arrow;local m;
psi(x)+sum(a[m]*N(x,m),m=1..M);end proc;
Phi(0,M);Phi(1,M);
```

$$N := (x, m) \rightarrow x^m (1 - x)$$

$$\Phi := (x, M) \rightarrow \text{local } m; \psi(x) + \left(\sum_{m=1}^M a_m N(x, m) \right)$$

1
2

Видим, что аппроксимирующая функция удовлетворяет граничным условиям при любом наборе числовых параметров a_m .

Теперь разработаем процедуру метода взвешенных невязок в форме метода поточечной коллокации

```
> collocations:=proc(M,X,a)local K,f,m,l,k,res;
K:=array(1..M,1..M):f:=array(1..M):a:=array(1..M):
for l from 1 to M do
  f[l]:=phi(X[l])-psi(X[l],M):
  for m from 1 to M do
    K[l,m]:=evalf(N(X[l],m)):
  end do:
end do:
res:=linalg[linsolve](K,f):
for k from 1 to M do a[k]:=res[k]; end do:
end proc:
```

Итак, все необходимые процедуры и функции созданы, и можно переходить к построению аппроксимации.

Для графического отображения результатов подготовим график заданной функции, но пока отображать его не будем

```
> p:=plot(phi(x), x=0..1, color=black,
          legend=`заданная функция`):
```


Получим аппроксимацию, используя только одну точку коллокации. Результат отобразим на графике

```
> M:=1;
X:=array(1..M, [0.5]);
collocations(M, X, a):
Phi(x, M);
P1:=plot(Phi(x, M), x=0..1, color=black, style=point,
         legend=`1 - точечная коллокация`):
plots[display]({P1, p});
```

$$M := 1$$

$$X := [.5]$$

$$x + 1 + .8284271252x(1 - x)$$




Попытаемся улучшить результат, выбрав две точки коллокации

```
> M:=2;
X:=array(1..M, [0.25, 0.75]);
collocations(M, X, a):
Phi(x, M);
P2:=plot(Phi(x, M), x=0..1, color=black, style=point,
         legend=`2 - точечная коллокация`):
plots[display]({P2, p});
```


$$M := 2$$

$$X := [.25, .75]$$

$$x + 1 + .5977887067x(1-x) + .4394250667x^2(1-x)$$



Видим, что сходимость аппроксимации хорошая и можно ограничиться полученным результатом.

ПРИМЕЧАНИЕ

При разработке процедуры метода поточечной коллокации мы исходили из формул (8.7), которые получены заранее, так сказать “вручную”. На самом деле нет необходимости предварительно вычислять интегралы (8.5) и (8.6), затем формировать матрицу и правые части системы (8.4). Все это можно поручить Maple. Ниже приводится соответствующая процедура

```
> collocations_1:=proc(M,X,a) local m,res,R,eq,eqns,var;
a:=array(1..M):
eqns:={}:var:={}:
R:=phi(x)-Phi(x,M):
for m from 1 to M do
    eq[m]:=int(R*W(x,m),x=0..1)=0:
    eqns:=eqns union {eq[m]}:
    var:=var union {a[m]}:
end do:
res:=solve(eqns,var):assign(res):
end proc:
```

Здесь, как обычно, $\{W(x,m); m=1,2,\dots\}$ — множество линейно независимых весовых функций; в качестве таких функций, в соответствии с методом поточечной коллокации, мы должны выбрать дельта-функции Дирака

```
> W:=(x,m)->Dirac(x-X[m]);
```

$$W := (x, m) \rightarrow \text{Dirac}(x - X_m)$$


Проверим работу этой процедуры. Для начала ограничимся одной точкой коллокации

```
> M:=1;X:=array(1..M,[0.5]);
  collocations_1(M,X,a):Phi(x,M);
  P1_1:=plot(Phi(x,M),x=0..1,color=black,style=point,
  legend=`1 - точечная коллокация`);
  plots[display]({P1_1,p});
```

$$M := 1$$

$$X := [.5]$$

$$x + 1 + .8284271240x(1 - x)$$




Как и следовало ожидать, мы получили тот же самый результат. Улучшим его, выберем две точки коллокации

```
> M:=2;X:=array(1..M,[0.25,0.75]);
  collocations_1(M,X,a):Phi(x,M);
  P2_1:=plot(Phi(x,M),x=0..1,color=black,
  style=point,symbol=circle,
  legend=`2 - точечная коллокация`);
  plots[display]({P2_1,P1_1,p});
```

$$M := 2$$

$$X := [.25, .75]$$

$$x + 1 + .5977887067x(1 - x) + .4394250667x^2(1 - x)$$



Рассмотрим теперь применение метода Галеркина для аппроксимации заданной функции. Разработаем соответствующую процедуру:


```
> Galerkin:=proc(M,a) local m,res,R,eq,eqns,var;
  a:=array(1..M);
  eqns:={}:var:={}:
  R:=phi(x)-Phi(x,M);
  for m from 1 to M do
    eq[m]:=int(R*N(x,m),x=0..1)=0;
    eqns:=eqns union {eq[m]};
    var:=var union {a[m]};
  end do;
  res:=solve(eqns,var);assign(res);
end proc;
```

Используя эту процедуру, построим аппроксимацию заданной функции. Начнем с одноэлементной аппроксимации

```
> M:=1;
  Galerkin(M,a);
  Phi(x,M);
  G1:=plot(Phi(x,M),x=0..1,color=black,style=point,
  legend=`1 - элементная аппроксимация`);
  plots[display]({G1,p});
```

$$M := 1$$

$$x + 1 - \frac{5(48\pi + \pi^3 - 192)x(1-x)}{\pi^3}$$



Двухэлементная аппроксимация:


```

> M:=2;
Galerkin(M, a) :
Phi(x, M);
G2:=plot(Phi(x, M), x=0..1, color=black, style=point,
symbol=cross, legend='2 - элементная аппроксимация'):
plots[display]({G1, G2, p});
  
```

$$M := 2$$

$$x + 1 + \frac{(-40320 + 720 \pi^2 + \pi^4 + 10560 \pi) x (1 - x)}{\pi^4}$$

$$- 7 \frac{(-11520 + 240 \pi^2 + \pi^4 + 2880 \pi) x^2 (1 - x)}{\pi^4}$$



Видим, что и в этом случае достаточно ограничиться двухэлементной аппроксимацией.

Аппроксимация решений дифференциальных уравнений

Рассмотрим в области Ω дифференциальное уравнение, которое запишем так

$$A(\varphi) = L\varphi + p = 0, \quad (8.9)$$

где L — соответствующий *линейный* дифференциальный оператор, а p не зависит от функции φ .

Решение уравнения (8.9) должно удовлетворять соответствующим краевым условиям на границе Γ области Ω , которые запишем в виде

$$B(\varphi) = M\varphi + r = 0, \quad (8.10)$$

где M — соответствующий *линейный* дифференциальный оператор, а r не зависит от функции φ .

Следуя описанной выше методике, попытаемся построить аппроксимацию $\tilde{\varphi}$ для решения φ , используя, как и в (8.1), разложение

$$\varphi \approx \tilde{\varphi} = \psi + \sum_{m=1}^M a_m N_m.$$

Здесь функция ψ и базисные функции N_m выбраны таким образом, что

$$M\psi = -r, \quad MN_m = 0, \quad m = 1, 2, \dots, \text{ на } \Gamma$$

и поэтому φ автоматически удовлетворяет краевым условиям (8.10) при произвольных коэффициентах a_m . Мы будем предполагать, что базисные функции непрерывно дифференцируемы¹.

Так как построенное разложение удовлетворяет краевым условиям, то для получения аппроксимации искомой функции φ необходимо только гарантировать, что $\tilde{\varphi}$ — приближенное решение дифференциального уравнения (8.9). Подставляя $\tilde{\varphi}$ в это уравнение, получим невязку по области R_Ω , которую в силу линейности оператора L можно записать в виде

$$R_\Omega \equiv A(\tilde{\varphi}) \equiv L\tilde{\varphi} + p \equiv L\psi + \left(\sum_{m=1}^M a_m LN_m \right) + p.$$

Для получения приближенного равенства $R_\Omega = 0$ всюду в Ω можно непосредственно применить описанные выше методики. Это позволит найти аппроксимацию $\tilde{\varphi}$ искомой функции φ . Как и ранее, для этой цели можно воспользоваться методом взвешенных невязок, выбирая некоторую систему весовых функций $\{W_n, n = 1, 2, \dots\}$ и требуя, чтобы

$$\int_{\Omega} W_n R_\Omega d\Omega \equiv \int_{\Omega} W_n \left\{ L\psi + \left(\sum_{m=1}^M a_m LN_m \right) + p \right\} d\Omega = 0. \quad (8.11)$$

¹ Это требование может быть ослаблено и тогда мы приходим к более общим формулировкам методов конечных элементов или методов граничных элементов.

Так как общее число неизвестных равно M , то, применяя уравнение (8.11) для $n = 1, 2, \dots, M$, получим систему линейных алгебраических уравнений, которую можно записать в виде

$$\mathbf{Ka} = \mathbf{f},$$

где теперь

$$K_{nm} = \int_{\Omega} W_n L N_m d\Omega, \quad 1 \leq m, n \leq M,$$

$$f_n = -\int_{\Omega} W_n p d\Omega - \int_{\Omega} W_n L \psi d\Omega, \quad 1 \leq n \leq M.$$

Вычислив элементы матрицы и правой части уравнений и решив затем полученную систему, мы определим неизвестные a_m , $m = 1, 2, \dots, M$, и тем самым получим приближенное решение заданного дифференциального уравнения.

Можно опять рассмотреть различные выборы весовых функций; для демонстрации деталей обратимся к примеру.

Задача 8.2. В теории упругого кручения призматических стержней встречается уравнение

$$\frac{\partial^2 \varphi}{\partial x^2} + \frac{\partial^2 \varphi}{\partial y^2} = -2G\theta, \quad (8.12)$$

где G — модуль сдвига материала, θ — угол скручивания для каждого поперечного сечения стержня, φ — функция напряжений, такая, что $\varphi = 0$ на контуре поперечного сечения.

Крутящий момент T задается формулой

$$T = 2 \int_{\Omega} \varphi d\Omega,$$

а касательное напряжение по направлению n в сечении стержня может быть получено по формуле

$$\tau = \frac{\partial \varphi}{\partial n}.$$

Требуется найти решение уравнения (8.12) в прямоугольнике $-3 \leq x \leq 3$, $-2 \leq y \leq 2$ с нулевыми граничными условиями.

Решение. Очевидно, что решение симметрично относительно прямых $x = 0$, $y = 0$, и поэтому мы ограничимся использованием четных базисных функций, которые также обладают этим свойством симметрии. Например, можно выбрать следующую систему базисных функций

$$N_1 = \cos\left(\frac{\pi x}{6}\right) \cos\left(\frac{\pi y}{4}\right), \quad N_2 = \cos\left(\frac{3\pi x}{6}\right) \cos\left(\frac{\pi y}{4}\right), \quad N_3 = \cos\left(\frac{\pi x}{6}\right) \cos\left(\frac{3\pi y}{4}\right),$$

и так далее. Разработаем процедуру, вычисляющую базисные функции, удовлетворяющие однородным граничным условиям и указанному свойству симметрии

```
> Bas_func:=proc(M,N) local m;
  for m from 1 to M do
    N[2*m-1]:=cos(Pi*x/6)*cos((2*m-1)*Pi*y/4);
    N[2*m]:=cos((2*m+1)*Pi*x/6)*cos(Pi*y/4);
  end do;
end proc;
```

Аппроксимирующую функцию определим процедурой

```
> psi:=proc(M) option operator, arrow; local m;
  sum(a[m]*N[m], 'm'=1..M); end proc;
```

$$\psi := M \rightarrow \text{local } m; \sum_{m=1}^M a_m N_m$$

Крутящий момент будем вычислять с помощью следующей процедуры

```
> torsion:=proc(M) option operator, arrow;
  2*int(int(psi(M), x=-3..3), y=-2..2)
end proc;
```

$$\text{torsion} := M \rightarrow 2 \int_{-2}^2 \int_{-3}^3 \psi(M) dx dy$$

И, наконец, разработаем процедуру, формирующую уравнения метода взвешенных невязок в форме метода Галеркина

```
> Galerkin:=proc(M) local eq, m, R, eqns, var, res;
  R:=diff(psi(M), x$2)+diff(psi(M), y$2)+2;
  eqns:={}:var:={}:
  for m from 1 to M do
    eq[m]:=int(int(R*N[m], x=-3..3), y=-2..2)=0;
    eqns:=eqns union {eq[m]}:
    var:=var union {a[m]}:
  end do:
  res:=solve(eqns, var):assign(res):
  print(res);
end proc;
```

Теперь можно приступить к построению приближенного решения задачи. Сначала вычислим несколько базисных функций

```
> Bas_func(5, N) : eval(N);
table([1 = cos(1/6 * pi * x) cos(1/4 * pi * y), 2 = cos(1/2 * pi * x) cos(1/4 * pi * y), 3 = cos(1/6 * pi * x) cos(3/4 * pi * y),
4 = cos(5/6 * pi * x) cos(1/4 * pi * y), 5 = cos(1/6 * pi * x) cos(5/4 * pi * y), 6 = cos(7/6 * pi * x) cos(1/4 * pi * y),
7 = cos(1/6 * pi * x) cos(7/4 * pi * y), 8 = cos(3/2 * pi * x) cos(1/4 * pi * y), 9 = cos(1/6 * pi * x) cos(9/4 * pi * y),
10 = cos(11/6 * pi * x) cos(1/4 * pi * y)
])
```

Получим решение при трехэлементной аппроксимации

```
> M:=3; a:=array(1..M);
psi(M); Galerkin(M); Mt:=evalf(torsion(M));
M:=3
a:=array(1..3, [ ])
a1*cos(1/6 * pi * x) cos(1/4 * pi * y) + a2*cos(1/2 * pi * x) cos(1/4 * pi * y) + a3*cos(1/6 * pi * x) cos(3/4 * pi * y)
{a2 = -512/15 * 1/pi^4, a1 = 4608/13 * 1/pi^4, a3 = -1536/85 * 1/pi^4}
Mt:=74.26510174
```

Пятиэлементная аппроксимация дает

```
> M:=5; a:=array(1..M);
Galerkin(M); Mt:=evalf(torsion(M));
M:=5
a:=array(1..5, [ ])
{a2 = -512/15 * 1/pi^4, a4 = 4608/545 * 1/pi^4, a1 = 4608/13 * 1/pi^4, a5 = 4608/1145 * 1/pi^4, a3 = -1536/85 * 1/pi^4}
Mt:=74.76356044
```

Точное значение крутящего момента равно 76.4 [8].

Продemonстрируем еще одну возможность построения приближенного решения рассматриваемой задачи, а именно, метод наименьших квадратов. Обычно метод наименьших квадратов не рассматривается как метод взвешенных невязок, однако, его тоже можно отнести к этому классу методов. Стандартная формулировка метода наименьших квадратов состоит в требовании минимизировать интеграл от квадрата невязки по области Ω . В данном случае попытаемся минимизировать функционал

$$J(a_1, a_2, a_3, \dots, a_M) = \int_{\Omega} R_{\Omega}^2 d\Omega \equiv \int_{\Omega} \left\{ L\psi + \left(\sum_{m=1}^M a_m LN_m \right) + p \right\}^2 d\Omega,$$

потребовав, чтобы

$$\frac{\partial J}{\partial a_n} = 0, \quad n = 1, 2, \dots, M.$$

В результате получим систему уравнений

$$\int_{\Omega} R_{\Omega} \frac{\partial R_{\Omega}}{\partial a_n} d\Omega = 0, \quad n = 1, 2, \dots, M,$$

которая совпадает с системой уравнений метода взвешенных невязок, если выбирать весовые функции по правилу

$$W_n = \frac{\partial R_{\Omega}}{\partial a_n} = LN_n, \quad n = 1, 2, \dots, M,$$

Разработаем процедуру метода наименьших квадратов применительно к рассматриваемой задаче

```
> LSM:=proc (M) local eq,m,R,F,eqns,var,res;
R:=diff(psi(M),x$2)+diff(psi(M),y$2)+2:
F:=int(int(R^2,x=-3..3),y=-2..2):
eqns:={}:var:={}:
for m from 1 to M do
    eq[m]:=diff(F,a[m])=0:
    eqns:=eqns union {eq[m]}:
    var:=var union {a[m]}:
end do:
res:=solve(eqns,var):assign(res):
print(res):
end proc:
```

Приступим теперь к построению аппроксимаций. Трехэлементная аппроксимация методом наименьших квадратов дает:

```
> M:=3;a:=array(1..M);psi(M);LSM(M);Mt:=evalf(torsion(M));
```

```
M:=3
```

```
a:=array(1..3,[ ])
```

$$a_1 \cos\left(\frac{1}{6} \pi x\right) \cos\left(\frac{1}{4} \pi y\right) + a_2 \cos\left(\frac{1}{2} \pi x\right) \cos\left(\frac{1}{4} \pi y\right) + a_3 \cos\left(\frac{1}{6} \pi x\right) \cos\left(\frac{3}{4} \pi y\right)$$

$$\left\{ a_2 = -\frac{512}{15} \frac{1}{\pi^4}, a_3 = -\frac{1536}{85} \frac{1}{\pi^4}, a_1 = \frac{4608}{13} \frac{1}{\pi^4} \right\}$$

```
Mt := 74.26510174
```

Получили тот же результат, что и методом Галеркина (продумайте, почему?). Отметим, что, вообще говоря, метод наименьших квадратов не эквивалентен методу Галеркина.

Аппроксимация решений дифференциальных уравнений и краевых условий

В предыдущем параграфе мы показали, как можно приближенно решить дифференциальное уравнение, используя разложение по базисным функциям и строя аппроксимацию, тождественно удовлетворяющую всем граничным условиям задачи. Здесь мы ослабим требование, предъявляемое аппроксимирующей функции и, тем самым, расширим выбор возможных видов базисных функций.

Будем теперь считать, что разложение

$$\varphi \approx \tilde{\varphi} = \sum_{m=1}^M a_m N_m \quad (8.13)$$

не удовлетворяет априори некоторому краевому условию или всем краевым условиям задачи. Тогда к невязке по области

$$R_\Omega \equiv A(\tilde{\varphi}) \equiv L\tilde{\varphi} + p$$

добавляется невязка в граничных условиях

$$R_\Gamma \equiv B(\tilde{\varphi}) \equiv M\tilde{\varphi} + r.$$

Можно теперь попытаться уменьшить взвешенную сумму невязок на границе и по области, полагая

$$\int_{\Omega} W_n R_\Omega d\Omega + \int_{\Gamma} \bar{W}_n R_\Gamma d\Gamma = 0, \quad (8.14)$$

где, вообще говоря, весовые функции W_n и \bar{W}_n могут быть выбраны независимо.

Если соотношение (8.14) выполняется для достаточно большого числа произвольных функций W_n и \bar{W}_n , то аппроксимация $\tilde{\varphi}$ должна приближать точное решение φ . Положение не меняется, если W_n и \bar{W}_n связаны каким-либо образом.

Подлежащую решению систему уравнений можно опять записать в виде

$$\mathbf{Ka} = \mathbf{f},$$

где теперь

$$K_{nm} = \int_{\Omega} W_n L N_m d\Omega + \int_{\Gamma} \bar{W}_n M N_m d\Gamma,$$

$$f_n = - \int_{\Omega} W_n p d\Omega - \int_{\Gamma} \bar{W}_n r d\Gamma.$$

Для иллюстрации применения этого процесса снова рассмотрим задачу о кручении стержня прямоугольного поперечного сечения, но на этот раз используем разложение по базисным функциям, не удовлетворяющее априори всем граничным условиям. Как и ранее, ограничимся четными базисными функциями. В качестве базисных функций можно выбрать, например, такую систему функций:

$$N_1 = (4 - y^2), N_2 = (4 - y^2)x^2, N_3 = (4 - y^2)y^2,$$

$$N_4 = (4 - y^2)x^2y^2, N_5 = (4 - y^2)x^4,$$

и так далее.

Аппроксимирующую функцию определим процедурой

```
> psi:=proc(x,y,M,N) option operator, arrow; local m,n;
(4-y^2)*sum(sum(C[m,n]*x^(2*m)*y^(2*n),m=0..M),n=0..N)
end proc;
```

$$\psi := (x, y, M, N) \rightarrow \text{local } m, n; (4 - y^2) \left(\sum_{n=0}^N \left(\sum_{m=0}^M C_{m,n} x^{(2m)} y^{(2n)} \right) \right)$$

Очевидно, эта функция удовлетворяет граничным условиям при $y = \pm 2$ и не удовлетворяет граничным условиям при $x = \pm 3$. Эти последние граничные условия учитываются путем включения их в уравнения метода взвешенных невязок, как это сделано в уравнении (8.14). Разработаем процедуру метода взвешенных невязок в форме метода Галеркина, то есть в качестве весовых функций будем использовать базисные функции $W_n = N_n$, $\bar{W}_n = N_n|_{\Gamma}$

```
> WDM:=proc(M,N) local eq,m,n,R,eqns,var,res;
R:=diff(psi(x,y,M,N),x$2)+diff(psi(x,y,M,N),y$2)+2:
eqns:={}:var:={}:
for m from 0 to M do
```

```

for n from 0 to N do
  eq[m,n]:=int(int(R*x^(2*m)*y^(2*n),x=-3..3),y=-2..2)+
  int(psi(3,y,M,N)*(3)^(2*m)*y^(2*n),y=-2..2)+
  int(psi(-3,y,M,N)*(-3)^(2*m)*y^(2*n),y=-2..2)=0:
  eqns:=eqns union {eq[m,n]}:
  var:=var union {C[m,n]}:
end do:
end do:
res:=solve(eqns,var):assign(res):
print(res);
end proc:

```

Нам понадобится процедура для вычисления крутящего момента

```

> torsion:=proc(M,N)option operator,arrow;
  2*int(int(psi(x,y,M,N),x=-3..3),y=-2..2)
end proc;

```

$$torsion := (M, N) \rightarrow 2 \int_{-2}^2 \int_{-3}^3 \psi(x, y, M, N) dx dy$$

Теперь приступим к построению приближенного решения задачи. Двухэлементная аппроксимация дает

```

> M:=1;N:=0;C:=array(0..M,0..N);psi(x,y,M,N);
WDM(M,N);Mt:=evalf(torsion(M,N));

```

$$M := 1$$

$$N := 0$$

$$C := \text{array}(0..1, 0..0, [\])$$

$$(4 - y^2) (C_{0,0} + C_{1,0} x^2)$$

$$\left\{ C_{1,0} = \frac{-6}{43}, C_{0,0} = \frac{189}{215} \right\}$$

$$Mt := 58.93953488$$

шестиэлементная аппроксимация

```

> M:=2;N:=1;C:=array(0..M,0..N);psi(x,y,M,N);

```

```

WDM(M,N);Mt:=evalf(torsion(M,N));
p||M||N:=plot(psi(3,y,M,N),
y=-2..2,color=black,linestyle=3):
M:=2
N:=1
C:=array(0..2,0..1,[ ])
(4-y^2)(C_{0,0}+C_{1,0}x^2+C_{2,0}x^4+C_{0,1}y^2+C_{1,1}x^2y^2+C_{2,1}x^4y^2)
{C_{0,0}=\frac{85732515}{107495464} C_{0,1}=\frac{7238385}{429981856} C_{2,1}=\frac{1015525}{429981856} C_{1,0}=\frac{-2719325}{53747732} C_{1,1}=\frac{-1598135}{214990928}
C_{2,0}=\frac{-579145}{107495464}}
Mt:=74.84402222

```

двенадцатиэлементная аппроксимация

```

> M:=3;N:=2;C:=array(0..M,0..N);
WDM(M,N);Mt:=evalf(torsion(M,N));
p||M||N:=plot(psi(3,y,M,N),y=-2..2,color=black):
M:=3
N:=2
C:=array(0..3,0..2,[ ])
Mt:=75.11590969

```


Напомним, что точное значение крутящего момента равно 76.4.

Повышение точности представления краевого условия $\tilde{\varphi} = 0$ на прямых $x = \pm 3$ с увеличением числа используемых элементов иллюстрируется следующим графиком

```

> plots[display]({p21,p32});

```



Заинтересованный читатель может поэкспериментировать с повышением точности вычислений, увеличивая число используемых элементов.

Естественные краевые условия

Примеры, разобранные выше, показывают, что можно вычислить коэффициенты разложения (8.13) и, таким образом, построить приближенное решение без априорного удовлетворения краевым условиям.

Использование уравнения (8.14) может потребовать вычисления интегралов, включающих производные от функции $\tilde{\varphi}$ вдоль границ, что может оказаться затруднительным, если эти границы криволинейны или имеют какие-либо другие осложняющие особенности. Оказывается, можно обойти эти затруднения, если воспользоваться так называемой *слабой формулировкой* метода взвешенных невязок.

Рассмотрим уравнение метода взвешенных невязок (8.14). В этом уравнении первое слагаемое

$$\int_{\Omega} W_n R_{\Omega} d\Omega \equiv \int_{\Omega} W_n (L\tilde{\varphi} + p) d\Omega,$$

как правило, можно преобразовать к виду

$$\int_{\Omega} W_n L\tilde{\varphi} d\Omega \equiv \int_{\Omega} (CW_n)(D\tilde{\varphi}) d\Omega + \int_{\Gamma} W_n E\tilde{\varphi} d\Gamma, \quad (8.15)$$

где C , D и E – линейные дифференциальные операторы более низкого порядка, чем исходный оператор L .

После подстановки (8.15) в (8.14) надлежащим выбором граничных весовых функций \bar{W}_n можно добиться того, что последнее слагаемое в (8.15) и часть второго слагаемого в (8.14) взаимно уничтожатся и, таким образом, будет исключен интеграл, содержащий $\tilde{\varphi}$ или ее производные вдоль границы. Это будет возможно только для определенных краевых условий, называемых *естественными*. Полученная таким образом форма уравнений и называется *слабой формой* уравнений метода взвешенных невязок. В общем случае, то есть когда граничные условия не являются естественными, применение такой процедуры выгоды не дает, но она может оказаться полезной при задании на границе производной. Еще одно преимущество такого преобразования состоит в определенной симметрии соответствующих алгебраических уравнений, а

также с получением, вообще говоря, более низкого порядка производных базисных функций².

При применении многих приближенных методов нет необходимости подчинять базисные функции однородным естественным условиям — достаточно, чтобы удовлетворялись так называемые *главные условия*. Данное обстоятельство существенно облегчает выбор базисных функций. В связи с этим возникает вопрос: как отличать естественные условия от главных? Можно указать такой признак для некоторого класса дифференциальных операторов, охватывающего все практически достаточно важные случаи [6]. Пусть оператор A имеет вид

$$Au = \sum_{k=0}^s \sum_{\substack{i_1, i_2, \dots, i_k=1 \\ j_1, j_2, \dots, j_k=1}}^m \frac{\partial^k}{\partial x_{i_1} \partial x_{i_2} \dots \partial x_{i_k}} \left(A_{i_1, i_2, \dots, i_k}^{j_1, j_2, \dots, j_k}(P) \frac{\partial^k u}{\partial x_{j_1} \partial x_{j_2} \dots \partial x_{j_k}} \right)$$

так что его порядок равен $2s$. Естественными будут те однородные краевые условия, в которые входят производные от функции u порядка s и выше, а главными — те условия, которые содержат производные от функции u только до порядка $s - 1$. Так, например, для оператора Лапласа ($s = 1$) условие $u|_{\Gamma}$ — главное, а условие

$$\frac{\partial u}{\partial n} + \sigma u \Big|_{\Gamma} = 0, \quad \sigma(P) \geq 0, \quad P \in \Gamma,$$

является естественным (n — вектор единичной внешней нормали к границе Γ).

Упомянутый признак, позволяющий отличать главные условия от естественных, остается в силе и в случае, когда оператор A — матричный, то есть в случае системы дифференциальных уравнений.

Естественные краевые условия для двумерного стационарного уравнения теплопроводности

Для иллюстрации и пояснения процесса перехода к слабой формулировке рассмотрим двумерное стационарное уравнение теплопроводности

$$\frac{\partial}{\partial x} \left(k \frac{\partial \varphi}{\partial x} \right) + \frac{\partial}{\partial y} \left(k \frac{\partial \varphi}{\partial y} \right) + Q = 0 \quad (8.16)$$

с краевыми условиями

$$\varphi|_{\Gamma_{\varphi}} = \bar{\varphi}, \quad k \frac{\partial \varphi}{\partial n} \Big|_{\Gamma_q} = -\bar{q}, \quad (8.17)$$

где k — коэффициент теплопроводности, Q — интенсивность внутренних источников тепла, n — внешняя нормаль к границе $\Gamma = \Gamma_{\varphi} + \Gamma_q$.

² Эта последняя особенность является важной при построении численных методов конечных элементов, которые основываются, как правило, на слабых формулировках соответствующих задач.

Будем использовать разложение по базисным функциям

$$\varphi \approx \tilde{\varphi} = \psi + \sum_{m=1}^M a_m N_m, \quad (8.18)$$

где ψ и базисные функции выбираются таким образом, что $\varphi|_{\Gamma_\varphi} = \bar{\varphi}$, то есть $\psi|_{\Gamma_\varphi} = \bar{\varphi}$ и $N_m = 0$, $m = 1, 2, \dots, M$ на Γ_φ . Уравнения метода взвешенных невязок в форме

$$\int_{\Omega} \left[\frac{\partial}{\partial x} \left(k \frac{\partial \tilde{\varphi}}{\partial x} \right) + \frac{\partial}{\partial y} \left(k \frac{\partial \tilde{\varphi}}{\partial y} \right) + Q \right] W_n dx dy + \int_{\Gamma} \left(k \frac{\partial \tilde{\varphi}}{\partial n} + \bar{q} \right) \bar{W}_n d\Gamma = 0$$

в пределе при $M \rightarrow \infty$ обеспечивают выполнение, как дифференциального уравнения, так и граничных условий. Первый интеграл можно переписать в виде (8.15), используя хорошо известную формулу Грина, согласно которой для достаточно гладких функций α и β справедливы тождества

$$\int_{\Omega} \alpha \frac{\partial \beta}{\partial x} dx dy = - \int_{\Omega} \beta \frac{\partial \alpha}{\partial x} dx dy + \int_{\Gamma} \alpha \beta n_x d\Gamma,$$

$$\int_{\Omega} \alpha \frac{\partial \beta}{\partial y} dx dy = - \int_{\Omega} \beta \frac{\partial \alpha}{\partial y} dx dy + \int_{\Gamma} \alpha \beta n_y d\Gamma,$$

где n_x и n_y — направляющие косинусы внешней нормали к замкнутому контуру Γ , ограничивающему область Ω в плоскости (x, y) , а интегрирование вдоль Γ осуществляется в направлении против часовой стрелки. Используя эти тождества и замечая, что

$$n_x \frac{\partial \alpha}{\partial x} + n_y \frac{\partial \alpha}{\partial y} = \frac{\partial \alpha}{\partial n},$$

уравнения метода взвешенных невязок можно преобразовать к виду

$$\begin{aligned} & - \int_{\Omega} \left[\left(\frac{\partial W_n}{\partial x} \right) \left(k \frac{\partial \tilde{\varphi}}{\partial x} \right) + \left(\frac{\partial W_n}{\partial y} \right) \left(k \frac{\partial \tilde{\varphi}}{\partial y} \right) \right] dx dy + \int_{\Omega} W_n Q dx dy + \\ & + \int_{\Gamma_\varphi + \Gamma_q} \left(k \frac{\partial \tilde{\varphi}}{\partial n} \right) W_n d\Gamma + \int_{\Gamma_q} \left(k \frac{\partial \tilde{\varphi}}{\partial n} + \bar{q} \right) \bar{W}_n d\Gamma = 0. \end{aligned}$$

Выберем теперь весовые функции в соответствие с требованием

$$W_n|_{\Gamma_\varphi} = 0, \quad \bar{W}_n|_{\Gamma_q} = -W_n.$$

Тогда интегралы, включающие градиент от функции $\tilde{\varphi}$ на границе, взаимно уничтожаются, и аппроксимирующее соотношение принимает вид

$$\int_{\Omega} \left[\left(\frac{\partial W_n}{\partial x} \right) \left(k \frac{\partial \tilde{\varphi}}{\partial x} \right) + \left(\frac{\partial W_n}{\partial y} \right) \left(k \frac{\partial \tilde{\varphi}}{\partial y} \right) \right] dx dy - \int_{\Omega} W_n Q dx dy + \int_{\Gamma_q} W_n \bar{q} d\Gamma = 0. \quad (8.19)$$

Подставляя в это уравнение разложение по базисным функциям (8.18), опять приходим к стандартной системе линейных алгебраических уравнений

$$\mathbf{K}\mathbf{a} = \mathbf{f},$$

с элементами матрицы и правой части

$$K_{nm} = \int_{\Omega} \left[\left(\frac{\partial W_n}{\partial x} \right) \left(k \frac{\partial N_m}{\partial x} \right) + \left(\frac{\partial W_n}{\partial y} \right) \left(k \frac{\partial N_m}{\partial y} \right) \right] dx dy, \quad 1 \leq n, m \leq M,$$

$$f_n = \int_{\Omega} W_n Q dx dy - \int_{\Gamma_q} W_n \bar{q} d\Gamma -$$

$$- \int_{\Omega} \left[\left(\frac{\partial W_n}{\partial x} \right) \left(k \frac{\partial \psi}{\partial x} \right) + \left(\frac{\partial W_n}{\partial y} \right) \left(k \frac{\partial \psi}{\partial y} \right) \right] dx dy, \quad 1 \leq n \leq M.$$

Отметим, что применение метода Галеркина приводит к симметричной матрице \mathbf{K} . Заметим также, что если $\bar{q} = 0$, то указанные границы более не входят явно в соотношения (8.19). Этот факт подчеркивает естественность задания на границе градиента в такой формулировке.

Рассмотрим конкретный пример.

Задача 8.3. Требуется решить задачу стационарной теплопроводности в материале с коэффициентом теплопроводности $k = 1$, занимающим квадратную область $-1 \leq x \leq 1$, $-1 \leq y \leq 1$. На сторонах $y = \pm 1$ поддерживается нулевая температура, тогда как через стороны $x = \pm 1$ подается тепло со скоростью $\cos\left(\frac{\pi y}{2}\right)$ на единицу длины.

Описывающее этот процесс дифференциальное уравнение стационарной теплопроводности при отсутствии источников тепла является частным случаем уравнения (8.16) и имеет вид

$$\frac{\partial^2 \varphi}{\partial x^2} + \frac{\partial^2 \varphi}{\partial y^2} = 0$$

с краевыми условиями типа (8.17)

$$\varphi|_{y=\pm 1} = 0, \quad \frac{\partial \varphi}{\partial x} \Big|_{x=\pm 1} = \pm \cos\left(\frac{\pi y}{2}\right).$$

Решение. Отметим, что данная задача допускает точное решение. Поэтому с целью дальнейшего сравнения приближенного решения с точным, прежде чем строить приближенное решение задачи, получим ее точное решение. Это последнее будем строить методом Фурье разделения переменных.

Сначала зададим уравнение

```
> PDE := Diff(u(x, y), x$2) + Diff(u(x, y), y$2) = 0;
```

$$PDE := \left(\frac{\partial^2}{\partial x^2} u(x, y) \right) + \left(\frac{\partial^2}{\partial y^2} u(x, y) \right) = 0$$

Как мы знаем, для решения дифференциальных уравнений в частных производных в Maple имеется команда `pdsolve(PDE, f, HINT, INTEGRATE, build)`. Напомним, что здесь параметр `PDE` задает дифференциальное уравнение. Параметр `f` задает неизвестную функцию (необязательный параметр). Параметр `HINT` (необязательный) задает форму, в которой ищется решение (может принимать значения `HINT = `*``, `HINT = `+`` или любое алгебраическое выражение). Параметр `INTEGRATE` (необязательный) указывает на необходимость автоматического интегрирования набора обыкновенных дифференциальных уравнений. Параметр `build` (необязательный) указывает на необходимость попытки сформировать явное выражение для неизвестной функции.

Так, следующая команда осуществляет разделение переменных и приводит два обыкновенных дифференциальных уравнения, которые получаются в результате такого разделения переменных

```
> sol := pdsolve(PDE, HINT = `*`);
```

$$sol := (u(x, y) = _F1(x) _F2(y)) \&where \left[\left\{ \frac{\partial^2}{\partial x^2} _F1(x) = _c1 _F1(x), \frac{\partial^2}{\partial y^2} _F2(y) = -_c1 _F2(y) \right\} \right]$$

Мы можем сразу же получить и общие решения этих обыкновенных дифференциальных уравнений

```
> pdsolve(PDE, HINT = `*`, INTEGRATE);
```

```
(u(x, y) = \_F1(x) \_F2(y)) &where
```

$$\left[\left\{ \left\{ _F1(x) = _C1 e^{\sqrt{-c1} x} + _C2 e^{-\sqrt{-c1} x} \right\}, \left\{ _F2(y) = _C3 \sin(\sqrt{-c1} y) + _C4 \cos(\sqrt{-c1} y) \right\} \right\} \right]$$

Теперь, согласно общей теории, нам надо сформулировать соответствующую задачу Штурма – Лиувилля. У нас однородные условия по переменной y . Поэтому соответствующее уравнение Штурма – Лиувилля будет уравнением относительно функции $_F2(y)$; его общее решение уже получено. Остается только найти собственные значения и отвечающие им собственные функции. Удобно записать общее решение уравнения Штурма – Лиувилля в виде

```
> Y :=
```

```
y -> C1 * cos(sqrt(lambda) * (y-1)) + C2 * sin(sqrt(lambda) * (y-1));
```

$$Y := y \rightarrow C1 \cos(\sqrt{\lambda} (y-1)) + C2 \sin(\sqrt{\lambda} (y-1))$$

Проверим правильность выбора общего решения в такой форме

```
> simplify(diff(Y(y), y$2) + lambda * Y(y)) = 0;
```

$$0 = 0$$

Составим теперь систему уравнений по граничным условиям

$$> \text{eq1} := Y(-1) = 0; \text{eq2} := Y(1) = 0;$$

$$\text{eq1} := C1 \cos(2\sqrt{\lambda}) - C2 \sin(2\sqrt{\lambda}) = 0$$

$$\text{eq2} := C1 = 0$$

Таким образом, одна из произвольных констант сразу определилась (в этом преимущество выбранной формы общего решения). Находим теперь собственные значения (впрочем, для этого нет необходимости прибегать к компьютеру!)

$$> _EnvAllSolutions := \text{true} : \text{solve}(\sin(2*\sqrt{\lambda}), \lambda);$$

$$\frac{1}{4} \pi^2 _Z1 \sim^2$$

Итак, собственные функции задачи Штурма – Лиувилля найдены:

$$> Y := (y, n) \rightarrow \sin(n*\pi/2*(y-1));$$

$$Y := (y, n) \rightarrow \sin\left(\frac{1}{2} n \pi (y - 1)\right)$$

Воспользовавшись полученной ранее «подсказкой», выданной нам системой Maple, выпишем теперь общее решение второго обыкновенного дифференциального уравнения (по переменной x) в следующей форме

$$> X := x \rightarrow C1*\cosh(n*\pi/2*x) + C2*\sinh(n*\pi/2*x);$$

$$X := x \rightarrow C1 \cosh\left(\frac{1}{2} n \pi x\right) + C2 \sinh\left(\frac{1}{2} n \pi x\right)$$

Тут же проверим правильность этого решения

$$> \text{simplify}(\text{diff}(X(x), x^2) - (n*\pi/2)^2*X(x)) = 0;$$

$$0 = 0$$

Составим теперь формальный ряд для искомого решения и ограничимся отрезком этого ряда

$$> u := \text{proc}(x, y, N) \text{ option operator, arrow; local } n;$$

$$\text{Sum}((C1[n]*\cosh(1/2*n*\pi*x) + C2[n]*\sinh(1/2*n*\pi*x)) * \sin(n*\pi/2*(y-1)), 'n' = 1..N);$$

$$\text{end proc};$$

$$\text{end proc};$$

$$u := (x, y, N) \rightarrow \text{local } n; \sum_{n'=1}^N \left(C1_{n'} \cosh\left(\frac{1}{2} n' \pi x\right) + C2_{n'} \sinh\left(\frac{1}{2} n' \pi x\right) \right) \sin\left(\frac{1}{2} n' \pi (y - 1)\right)$$

Для определения произвольных постоянных нам понадобится производная решения по x , которую мы и определим

$$> u1 := \text{diff}(u(x, y, N), x); u1 := \text{unapply}(u1, x, y, N);$$

$$uI := \sum_{n=1}^N \left(\frac{1}{2} C1_n \sinh\left(\frac{1}{2} n \pi x\right) n \pi + \frac{1}{2} C2_n \cosh\left(\frac{1}{2} n \pi x\right) n \pi \right) \sin\left(\frac{1}{2} n \pi (y-1)\right)$$

$$uI := (x, y, N) \rightarrow \sum_{n=1}^N \left(\frac{1}{2} C1_n \sinh\left(\frac{1}{2} n \pi x\right) n \pi + \frac{1}{2} C2_n \cosh\left(\frac{1}{2} n \pi x\right) n \pi \right) \sin\left(\frac{1}{2} n \pi (y-1)\right)$$

Вычислим теперь значения производной в точках $x = \pm 1$

> u1L:=u1(-1, y, N); u1R:=u1(1, y, N);

$$u1L := \sum_{n=1}^N \left(-\frac{1}{2} C1_n \sinh\left(\frac{1}{2} n \pi\right) n \pi + \frac{1}{2} C2_n \cosh\left(\frac{1}{2} n \pi\right) n \pi \right) \sin\left(\frac{1}{2} n \pi (y-1)\right)$$

$$u1R := \sum_{n=1}^N \left(\frac{1}{2} C1_n \sinh\left(\frac{1}{2} n \pi\right) n \pi + \frac{1}{2} C2_n \cosh\left(\frac{1}{2} n \pi\right) n \pi \right) \sin\left(\frac{1}{2} n \pi (y-1)\right)$$

Теперь нам надо разложить в ряд Фурье граничную функцию. Подсчитаем коэффициенты Фурье этой функции: для $n \neq 1$

> assume(n, integer):

h:=Int(cos(1/2*Pi*y)*sin(1/2*n*Pi*(y-1)), y = -1 .. 1);

h:=value(h);

Norma:=Int(sin(1/2*n*Pi*(y-1))^2, y = -1 .. 1);

Norma:=value(Norma);

$$h := \int_{-1}^1 \cos\left(\frac{1}{2} \pi y\right) \sin\left(\frac{1}{2} n \pi (y-1)\right) dy$$

h := 0

$$Norma := \int_{-1}^1 \sin\left(\frac{1}{2} n \pi (y-1)\right)^2 dy$$

Norma := 1

для $n = 1$

> h1:=Int(cos(1/2*Pi*y)*sin(1/2*Pi*(y-1)), y = -1 .. 1);

h1:=value(h1);

Normal:=Int(sin(1/2*Pi*(y-1))^2, y = -1 .. 1);

Normal:=value(Normal);

$$h1 := \int_{-1}^1 \cos\left(\frac{1}{2} \pi y\right) \sin\left(\frac{1}{2} \pi (y-1)\right) dy$$

$$h1 := -1$$

$$Normal := \int_{-1}^1 \sin\left(\frac{1}{2}\pi(y-1)\right)^2 dy$$

$$Normal := 1$$

Таким образом, из всей суммы (решения $u(x, y)$) остается только одно первое слагаемое (при $n = 1$): система уравнений для определения констант при $n \neq 1$ имеет только тривиальное решение. Система уравнений для определения констант при $n = 1$ имеет вид

$$\begin{aligned} > E1 := 1/2 * C1 * \sinh(-1/2 * \pi) * \pi + 1/2 * C2 * \cosh(-1/2 * \pi) * \pi = \\ & -h1 / Normal; \\ E2 := 1/2 * C1 * \sinh(1/2 * \pi) * \pi + 1/2 * C2 * \cosh(1/2 * \pi) * \pi = \\ & h1 / Normal; \end{aligned}$$

$$E1 := -\frac{1}{2} C1 \sinh\left(\frac{1}{2}\pi\right) \pi + \frac{1}{2} C2 \cosh\left(\frac{1}{2}\pi\right) \pi = 1$$

$$E2 := \frac{1}{2} C1 \sinh\left(\frac{1}{2}\pi\right) \pi + \frac{1}{2} C2 \cosh\left(\frac{1}{2}\pi\right) \pi = -1$$

Решение этой системы

$$\begin{aligned} > res := \text{solve}(\{E1, E2\}, \{C1, C2\}); \text{assign}(res); \\ res := \{C2 = 0, C1 = -4 \frac{e^{(1/2)\pi}}{\pi(-1 + e^\pi)}\} \end{aligned}$$

Легко видеть, что константа $C1$ на самом деле равна

$$> C_1 := -2 / \pi / \sinh(1/2 * \pi);$$

$$C_1 := -2 \frac{1}{\pi \sinh\left(\frac{1}{2}\pi\right)}$$

Проверка этого факта:

$$\begin{aligned} > \text{simplify}(\text{convert}(C1 - C_1, \text{exp})); \\ 0 \end{aligned}$$

Таким образом, окончательно получаем решение в виде

$$> u := C_1 * \cosh(1/2 * \pi * x) * \sin(1/2 * \pi * (y - 1));$$

$$u := -2 \frac{\cosh\left(\frac{1}{2}\pi x\right) \sin\left(\frac{1}{2}\pi(y-1)\right)}{\pi \sinh\left(\frac{1}{2}\pi\right)}$$

Это решение оформим в виде функции

> u:=unapply(u, x, y);

$$u := (x, y) \rightarrow -2 \frac{\cosh\left(\frac{1}{2} \pi x\right) \sin\left(\frac{1}{2} \pi (y-1)\right)}{\pi \sinh\left(\frac{1}{2} \pi\right)}$$

Проверка полученного решения: уравнение Лапласа

> linalg[laplacian](u(x, y), [x, y]);

0

граничные условия при $y = \pm 1$

> u(x, -1); u(x, 1);

0

0

граничные условия при $x = \pm 1$

> u1:=unapply(diff(u(x, y), x), x, y);

expand(u1(-1, y)); expand(u1(1, y));

$$u1 := (x, y) \rightarrow - \frac{\sinh\left(\frac{1}{2} \pi x\right) \sin\left(\frac{1}{2} \pi (y-1)\right)}{\sinh\left(\frac{1}{2} \pi\right)} - \cos\left(\frac{1}{2} \pi y\right) \cos\left(\frac{1}{2} \pi y\right)$$

Рассмотрим теперь построение приближенного решения нашей задачи. С целью рассмотрения еще одного приближенного метода построим приближенное решение задачи методом Ритца. Метод Ритца представляет собой один из методов приближенного решения вариационных задач. Стандартная формулировка метода Ритца состоит в следующем [6]. Всякое уравнение вида $Au = f(P)$, где A — положительный оператор, сводится к нахождению минимума функционала

$$F(u) = (Au, u) - 2(f, u) \equiv \int_{\Omega} uAud\Omega - 2 \int_{\Omega} ufd\Omega. \quad (8.20)$$

Для отыскания минимума функционала (8.20) выберем последовательность базисных функций $N_m(P)$, $m = 1, 2, \dots$, принадлежащих области определения оператора A (это означает, помимо соответствующих требований гладкости, также и то, что функции удовлетворяют некоторым, вполне определенным, граничным условиям).

Кроме того, эти функции должны быть линейно независимы и при любом n должны удовлетворять условию полноты.

Построим линейную комбинацию первых n базисных функций с произвольными числовыми коэффициентами a_m

$$u_M = \sum_{m=1}^M a_m N_m(P), P \in \Omega.$$

Подставим $u_M(P)$ вместо $u(P)$ в функционал (8.20); это превратит $F(u)$ в функцию n независимых переменных a_m :

$$F(u_M) = \sum_{j,k=1}^M a_j a_k \left(\int_{\Omega} N_k A N_j d\Omega \right) - 2 \sum_{j=1}^M a_j \left(\int_{\Omega} N_j f d\Omega \right). \quad (8.21)$$

Выберем коэффициенты a_m так, чтобы функция (8.21) приняла минимальное значение. Функция (8.21) достигает минимума при тех значениях независимых переменных, которые обращают в нуль ее первые производные:

$$\frac{\partial F(u_M)}{\partial a_i} = 0, i = 1, 2, \dots, M. \quad (8.22)$$

Уравнения (8.22) являются, как известно, лишь необходимыми условиями минимума функции $F(u_M)$, однако положительность оператора A является как раз тем условием, которое обеспечивает минимум функции $F(u_M)$, если коэффициенты a_m удовлетворяют системе (8.22).

Выполнив дифференцирование в (8.22), приходим к следующей алгебраической системе уравнений относительно коэффициентов a_m

$$\sum_{k=1}^M a_k \left(\int_{\Omega} N_k A N_j d\Omega \right) = \left(\int_{\Omega} N_j f d\Omega \right), j = 1, 2, \dots, M. \quad (8.23)$$

Система уравнений Ритца (8.23) всегда разрешима, если оператор A положительный.

Отметим, что система (8.23) по виду тождественна с системой уравнений, к которой приводит метод Галеркина. Отсюда нетрудно заключить, что методы Галеркина и Ритца совпадают, если оператор A положительный (более того, он должен быть положительно определенным [6]). В общем же случае метод Ритца неприменим, тогда как метод Галеркина сохраняет силу.

Итак, чтобы решить краевую задачу методом Ритца сначала надо свести ее к соответствующей вариационной задаче. Общая формулировка в форме (8.20) часто требует дополнительных преобразований (как правило, интегрирований по частям или применение формул Грина), подобных тому, как это сделано в формуле (8.15).

Рассмотрим теперь решение нашей задачи методом Ритца в системе Maple. Отметим, что оператор Лапласа с краевыми условиями первого, второго или третьего рода является положительно определенным [6], поэтому применение метода Ритца должно

привести к успеху. Заметим далее, что соответствующая вариационная формулировка для уравнения Лапласа с граничными условиями второго рода сводится к отысканию минимума функционала [6]

$$F(\varphi) = \int_{\Omega} (\mathbf{grad}\varphi)^2 d\Omega - 2 \int_{\Gamma} \varphi \bar{q} d\Gamma$$

на множестве функций, которые никаким краевым условиям не подчинены (условие второго рода — естественное условие для оператора Лапласа).

Зададим аппроксимирующую функцию по формуле:

```
> phi:=proc(x,y,M,N) option operator, arrow; local i,j;
  (1-y^2)*sum(sum(a[i,j]*x^(2*i)*y^(2*j),i=0..M),j=0..N);
end proc;
```

$$\phi := (x, y, M, N) \rightarrow \text{local } i, j; (1 - y^2) \left(\sum_{j=0}^N \left(\sum_{i=0}^M a_{i,j} x^{(2i)} y^{(2j)} \right) \right)$$

Эта функция автоматически удовлетворяет условиям на прямых $y = \pm 1$, но не удовлетворяет условиям на прямых $x = \pm 1$; последние являются естественными условиями. Разработаем процедуру метода Рунге:

```
> SOL:=proc(M,N) local F,eqns,var,eq,i,j,res;
  F:=simplify(
  int(int(Dphi_x(x,y,M,N)^2+Dphi_y(x,y,M,N)^2,
  y=-1..1),x=-1..1)
  -2*int(phi(1,y,M,N)*cos(Pi*y/2),y=-1..1)
  -2*int(phi(-1,y,M,N)*cos(Pi*y/2),y=-1..1));
  eqns:={}:var:={}:
  for i from 0 to M do
    for j from 0 to N do
      var:=var union {a[i,j]}:
      eq[i,j]:=diff(F,a[i,j])=0:
      eqns:=eqns union {eq[i,j]}:
    od;
  od;
  res:=solve(eqns,var);
  assign(res);
end proc;
```

Далее, нам необходимы производные — компоненты градиента


```

> dx:=simplify(diff(phi(x,y,M,N),x)):
dy:=simplify(diff(phi(x,y,M,N),y)):
Dphi_x:=unapply(dx,x,y,M,N);Dphi_y:=unapply(dy,x,y,M,N);

```

$$Dphi_x := (x, y, M, N) \rightarrow -2(-1+y^2) \left(\sum_{j=0}^N y^{(2j)} \left(\sum_{i=0}^M i x^{(2i-1)} a_{i,j} \right) \right)$$

$Dphi_y := (x, y, M, N) \rightarrow$

$$-2y \left(\sum_{j=0}^N y^{(2j)} \left(\sum_{i=0}^M a_{i,j} x^{(2i)} \right) \right) + 2 \left(\sum_{j=0}^N \left(\sum_{i=0}^M a_{i,j} x^{(2i)} \right) j y^{(2j-1)} \right) - 2 \left(\sum_{j=0}^N \left(\sum_{i=0}^M a_{i,j} x^{(2i)} \right) j y^{(2j-1)} \right) y^2$$

Теперь можно приступить к построению приближенных решений.

Двухэлементная аппроксимация

```

> M:=1;N:=0;a:=array(0..M,0..N);i:='i':j:='j':
phi(x,y,M,N);Dphi_x(x,y,M,N);Dphi_y(x,y,M,N);
SOL(M,N);
py_|M|N:=
plot(Dphi_x(1,y,M,N),y=0..1,color=black,linestyle=3,
legend=`2 - элементная аппроксимация`):

```

$$M := 1$$

$$N := 0$$

$$a := \text{array}(0..1, 0..0, [\])$$

$$(1 - y^2) (a_{0,0} + a_{1,0} x^2)$$

$$-2(-1 + y^2) x a_{1,0}$$

$$-2y (a_{0,0} + a_{1,0} x^2)$$

$$\left\{ a_{0,0} = \frac{54}{7} \frac{1}{\pi^3}, a_{1,0} = \frac{90}{7} \frac{1}{\pi^3} \right\}$$

четырёхэлементная аппроксимация

```

> M:=1;N:=1;a:=array(0..M,0..N);i:='i':j:='j':
phi(x,y,M,N);Dphi_x(x,y,M,N);Dphi_y(x,y,M,N);
SOL(M,N);
py_|M|N:=
plot(Dphi_x(1,y,M,N),y=0..1,color=black,
style=point,symbol=circle,
legend=`4 - элементная аппроксимация`):

```

$$\begin{aligned}
M &:= 1 \\
N &:= 1 \\
a &:= \text{array}(0..1, 0..1, []) \\
(1-y^2) &(a_{0,0} + a_{1,0}x^2 + a_{0,1}y^2 + a_{1,1}x^2y^2) \\
&- 2(-1+y^2)(xa_{1,0} + y^2xa_{1,1}) \\
&- 2y(a_{0,0} + a_{1,0}x^2 + y^2(a_{0,1} + a_{1,1}x^2)) + 2(a_{0,1} + a_{1,1}x^2)y - 2(a_{0,1} + a_{1,1}x^2)y^3 \\
\{a_{0,1} = &-\frac{105541\pi^2 - 5260}{472\pi^5}, a_{0,0} = \frac{15491\pi^2 - 2324}{472\pi^5}, a_{1,1} = \frac{472599\pi^2 - 980}{472\pi^5}, a_{1,0} = -\frac{157541\pi^2 - 444}{472\pi^5}\}
\end{aligned}$$

шестиэлементная аппроксимация

```

> M:=2;N:=1;a:=array(0..M,0..N);i:='i':j:='j':
  phi(x,y,M,N);Dphi_x(x,y,M,N);Dphi_y(x,y,M,N);
  SOL(M,N);
  py_|M|N:=
  plot(Dphi_x(1,y,M,N),y=0..1,color=black,
  style=point,symbol=cross,
  legend=`б - элементная аппроксимация`):

```


$$\begin{aligned}
M &:= 2 \\
N &:= 1 \\
a &:= \text{array}(0..2, 0..1, []) \\
(1-y^2) &(a_{0,0} + a_{1,0}x^2 + a_{2,0}x^4 + a_{0,1}y^2 + a_{1,1}x^2y^2 + a_{2,1}x^4y^2) \\
&- 2(-1+y^2)(xa_{1,0} + 2x^3a_{2,0} + y^2(xa_{1,1} + 2x^3a_{2,1})) \\
&- 2y(a_{0,0} + a_{1,0}x^2 + a_{2,0}x^4 + y^2(a_{0,1} + a_{1,1}x^2 + a_{2,1}x^4)) + 2(a_{0,1} + a_{1,1}x^2 + a_{2,1}x^4)y \\
&- 2(a_{0,1} + a_{1,1}x^2 + a_{2,1}x^4)y^3 \\
\{a_{2,1} = &\frac{14175397\pi^2 - 3920}{4472\pi^5}, a_{0,1} = \frac{525347\pi^2 - 3584}{4472\pi^5}, a_{1,1} = -\frac{47259\pi^2 - 88}{172\pi^5}, a_{0,0} = -\frac{52555\pi^2 - 1264}{4472\pi^5}, \\
&a_{2,0} = -\frac{4725163\pi^2 - 1632}{4472\pi^5}, a_{1,0} = \frac{22525\pi^2 - 168}{172\pi^5}\}
\end{aligned}$$

Сравнение нормальных производных точного и приближенного решений на границе $x = 1$ представим в виде графиков:

```

> py2:=plot(u1(1,y),y=0..1,color=black,
  legend=`Точное решение`):
  plots[display]({py2,py_10,py_11,py_21});

```



Методы граничного решения

До сих пор для построения приближенных решений дифференциальных уравнений с соответствующими граничными условиями мы использовали системы базисных функций, такие, что

1. либо аппроксимация полностью удовлетворяла краевым условиям, но не удовлетворяла априори дифференциальному уравнению в области Ω ,
2. либо аппроксимация ни полностью, ни частично не удовлетворяла краевым условиям, как не удовлетворяла и дифференциальному уравнению.

Очевидно, существует и третья возможность, когда базисные функции таковы, что аппроксимация автоматически удовлетворяет дифференциальному уравнению, но не удовлетворяет непосредственно граничным условиям. Если рассматриваемое дифференциальное уравнение линейно, то такая возможность может быть реализована выбором базисных функций, которые сами удовлетворяют этому дифференциальному уравнению. Выбирая так базисные функции, запишем

$$\varphi \approx \tilde{\varphi} = \sum_{m=1}^M a_m N_m.$$

Так как невязка по области равна нулю ($R_\Omega = 0$), то уравнение метода взвешенных невязок сводится к соотношению

$$\int_{\Gamma} \bar{W}_n R_\Gamma d\Gamma = 0, \quad n = 1, 2, 3, \dots, M.$$

Теперь достаточно определить систему весовых функций \bar{W}_n , и, причем только лишь на границе Γ . Следует отметить, что такую систему функций выбрать труднее. Здесь мы рассмотрим выбор подходящих решений для уравнения Лапласа, для которого такой выбор прост. Как известно, если имеется некоторая аналитическая функция комплексной переменной z , то ее вещественная и мнимая части должны автоматически удовлетворять уравнению Лапласа. Этот факт позволяет непосредственно использовать аналитическую функцию $f(z) = z^n$ для получения системы функций, удовлетворяющих уравнению Лапласа.

Продemonстрируем применение метода граничного решения на примере повторного решения задачи кручения стержня, рассматривавшейся ранее.

Задача кручения описывается дифференциальным уравнением (постоянный множитель $G\theta$ для удобства вычислений внесем в неизвестную функцию)

$$\frac{\partial^2 \varphi}{\partial x^2} + \frac{\partial^2 \varphi}{\partial y^2} = -2, \quad (8.24)$$

в прямоугольной области $-3 \leq x \leq 3$, $-2 \leq y \leq 2$ с нулевыми граничными условиями.

Чтобы иметь возможность воспользоваться решениями уравнения Лапласа сведем уравнение Пуассона (8.24) к уравнению Лапласа, введя новую неизвестную функцию u по формуле $\varphi = u - \frac{x^2 + y^2}{2}$. Тогда задача кручения сводится к решению уравнения Лапласа

$$\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} = 0, \quad -3 \leq x \leq 3, \quad -2 \leq y \leq 2,$$

с условием $u = \frac{x^2 + y^2}{2}$ на границе.

Построим первые десять базисных функций, удовлетворяющих уравнению Лапласа

```
> MM:=10:
```

```
for m from 0 to MM do
```

```
N[m]:=unapply(evalc(Re((x+I*y)^m)),x,y)
```

```
end do;
```

$$N_0 := 1$$

$$N_1 := (x, y) \rightarrow x$$

$$N_2 := (x, y) \rightarrow x^2 - y^2$$

$$N_3 := (x, y) \rightarrow x^3 - 3xy^2$$

$$N_4 := (x, y) \rightarrow x^4 - 6x^2y^2 + y^4$$

$$N_5 := (x, y) \rightarrow x^5 - 10x^3y^2 + 5xy^4$$

$$N_6 := (x, y) \rightarrow x^6 - 15x^4y^2 + 15x^2y^4 - y^6$$

$$N_7 := (x, y) \rightarrow x^7 - 21x^5y^2 + 35x^3y^4 - 7xy^6$$

$$N_8 := (x, y) \rightarrow x^8 - 28x^6y^2 + 70x^4y^4 - 28x^2y^6 + y^8$$

$$N_9 := (x, y) \rightarrow x^9 - 36x^7y^2 + 126x^5y^4 - 84x^3y^6 + 9xy^8$$

$$N_{10} := (x, y) \rightarrow x^{10} - 45x^8y^2 + 210x^6y^4 - 210x^4y^6 + 45x^2y^8 - y^{10}$$

Определим аппроксимацию с учетом симметрии по формулам

```
> u:=proc(x,y,M) option operator, arrow; local m;
  sum(a[m]*N[2*m](x,y), m=0..M)
end proc;
phi:=(x,y,M) -> u(x,y,M) - (x^2+y^2)/2;
```

$$u := (x, y, M) \rightarrow \text{local } m; \sum_{m=0}^M a_m N_{2m}(x, y)$$

$$\phi := (x, y, M) \rightarrow u(x, y, M) - \frac{1}{2}x^2 - \frac{1}{2}y^2$$

Разработаем процедуру для вычисления крутящего момента

```
> torsion:=proc(M) option operator, arrow;
  2*int(int(u(x,y,M) - (x^2+y^2)/2, x=-3..3), y=-2..2)
end proc;
```

$$\text{torsion} := M \rightarrow 2 \int_{-2}^2 \int_{-3}^3 u(x, y, M) - \frac{1}{2}x^2 - \frac{1}{2}y^2 dx dy$$

И, наконец, разработаем процедуру, формирующую уравнения метода взвешенных невязок в форме метода граничного решения

```
> BSM:=proc(M) local eq,m,eqns,var,res;
  eqns:={}:var:={}:
  for m from 0 to M do
    eq[m]:=int((u(3,y,M) - (9.+y^2)/2)*N[2*m](3,y), y=0..2) +
      int((u(x,2,M) - (x^2+4.)/2)*N[2*m](x,2), x=0..3)=0:
    eqns:=eqns union {eq[m]}:
    var:=var union {a[m]}:
  end do:
  res:=solve(eqns,var):assign(res):
  print(res);
end proc:
```

Теперь можно приступить к построению аппроксимаций. Одноэлементная аппроксимация:

```
> M:=0;a:=array(0..M);u(x,y,M);
```

```
BSM(M);Mt:=evalf(torsion(M));
M:=0
a:=array(0..0,[ ])
a0
{a0=4.166666666}
Mt:=96.
```

двухэлементная аппроксимация

```
> M:=1;a:=array(0..M);u(x,y,M);
BSM(M);Mt:=evalf(torsion(M));
M:=1
a:=array(0..1,[ ])
a0 + a1(x2 - y2)
{a1=.2331264548, a0=3.591621411}
Mt:=87.04794410
```


трехэлементная аппроксимация

```
> M:=2;a:=array(0..M);u(x,y,M);
BSM(M);Mt:=evalf(torsion(M));
M:=2
a:=array(0..2,[ ])
a0 + a1(x2 - y2) + a2(x4 - 6x2y2 + y4)
{a1=.2749006572, a2=-.01437890550, a0=3.215379174}
Mt:=75.50511528
```

Ограничимся полученным решением (напомним, что точное значение крутящего момента равно 76.4)

Посмотрим еще на поведение трехэлементной аппроксимации на прямых $x = 3$, $y = 2$

```
> plot(phi(3,y,M),y=0..2,color=black,
title=`Поведение аппроксимации на прямой x = 3`);
plot(phi(x,2,M),x=0..3,color=black,
title=`Поведение аппроксимации на прямой y = 2`);
```



Отметим, что метод взвешенных невязок может быть применен также к решению систем дифференциальных уравнений и к решению нелинейных задач.

Нелинейные задачи

В этом случае применение стандартной процедуры метода взвешенных невязок приводит к системе нелинейных уравнений. Система нелинейных уравнений может быть решена с помощью подходящего итерационного метода.

Рассмотрим соответствующие построения на примере задачи двумерного стационарного распространения тепла в материале, теплопроводность которого зависит от температуры. Описывающее этот процесс дифференциальное уравнение имеет вид

$$\frac{\partial}{\partial x} \left(k(\varphi) \frac{\partial \varphi}{\partial x} \right) + \frac{\partial}{\partial y} \left(k(\varphi) \frac{\partial \varphi}{\partial y} \right) + Q = 0,$$

где $k(\varphi)$ — заданная функция. Пусть на части границы Γ_φ задано условие Дирихле $\varphi = \bar{\varphi}$, а на части границы Γ_q — условие Неймана $k(\varphi) \partial \varphi / \partial n = -\bar{q}$.

Будем строить аппроксимацию в форме

$$\tilde{\varphi} = \psi + \sum_{m=1}^M a_m N_m,$$

где ψ и N_m , как обычно, выбраны так, чтобы автоматически обеспечить выполнение главного краевого условия на Γ_φ . Используя соотношения (11.19), приходим к системе уравнений метода взвешенных невязок

$$\int_{\Omega} \left[\left(\frac{\partial W_n}{\partial x} \right) \left(k(\tilde{\varphi}) \frac{\partial \tilde{\varphi}}{\partial x} \right) + \left(\frac{\partial W_n}{\partial y} \right) \left(k(\tilde{\varphi}) \frac{\partial \tilde{\varphi}}{\partial y} \right) \right] dx dy -$$

$$- \int_{\Omega} W_n Q dx dy + \int_{\Gamma_q} W_n \bar{q} d\Gamma = 0, \quad n = 1, 2, \dots, M.$$

Эта система нелинейных уравнений требует итерационного решения. Следуя методу простой итерации, можно исходить из некоторого начального приближения

$$\mathbf{a} = \mathbf{a}^0 = (a_1^0, a_2^0, \dots, a_M^0)^T$$

и отвечающего ему приближенного решения $\tilde{\varphi}^0$, а затем получить улучшенное решение \mathbf{a}^1 из линейного уравнения

$$\int_{\Omega} \left[\left(\frac{\partial W_n}{\partial x} \right) \left(k(\tilde{\varphi}^0) \frac{\partial \tilde{\varphi}^1}{\partial x} \right) + \left(\frac{\partial W_n}{\partial y} \right) \left(k(\tilde{\varphi}^0) \frac{\partial \tilde{\varphi}^1}{\partial y} \right) \right] dx dy -$$

$$- \int_{\Omega} W_n Q dx dy + \int_{\Gamma_q} W_n \bar{q} d\Gamma = 0, \quad n = 1, 2, \dots, M.$$

Общая итерационная схема

$$\int_{\Omega} \left[\left(\frac{\partial W_n}{\partial x} \right) \left(k(\tilde{\varphi}^{l-1}) \frac{\partial \tilde{\varphi}^l}{\partial x} \right) + \left(\frac{\partial W_n}{\partial y} \right) \left(k(\tilde{\varphi}^{l-1}) \frac{\partial \tilde{\varphi}^l}{\partial y} \right) \right] dx dy -$$

$$- \int_{\Omega} W_n Q dx dy + \int_{\Gamma_q} W_n \bar{q} d\Gamma = 0, \quad n = 1, 2, \dots, M.$$

применяется до тех пор, пока процесс не сойдется в пределах заданной точности приближения.

Задача 8.4. Рассмотрим одномерную задачу теплопроводности с генерированием тепла в области, где теплопроводность зависит от температуры: задача описывается уравнением

$$\frac{d}{dx} \left(k(\varphi) \frac{d\varphi}{dx} \right) = -10x$$

с граничными условиями $\varphi = 0$ для $x = 0$ и $x = 1$ и $k = 1 + 0,1\varphi$.

Решение. Решение этой задачи методом взвешенных невязок может быть получено путем выполнения описанных выше шагов. Например, для системы базисных функций

$$N_m = x^m (1 - x), \quad m = 1, 2, \dots$$

разложение

$$\tilde{\varphi} = \sum_{m=1}^M a_m x^m (1-x)$$

автоматически удовлетворяет краевым условиям при $x = 0$ и $x = 1$.

Применяя метод взвешенных невязок, получаем систему уравнений

$$\int_0^1 \left\{ \frac{d}{dx} \left[k(\tilde{\varphi}) \frac{d\tilde{\varphi}}{dx} \right] + 10x \right\} W_n dx = 0, \quad n = 1, 2, \dots, M.$$

Эту систему будем решать методом итераций по схеме

$$\int_0^1 \left\{ \frac{d}{dx} \left[k(\tilde{\varphi}^{l-1}) \frac{d\tilde{\varphi}^l}{dx} \right] + 10x \right\} W_n dx = 0, \quad n = 1, 2, \dots, M, \quad l = 1, 2, \dots,$$

исходя из начального приближения $\tilde{\varphi}^0 = 0$; в качестве весовых функций выберем базисные функции.

Разработаем необходимые процедуры. Базисные функции и аппроксимирующая функция:

```
> N := (x, m) -> x^m * (1-x);
phi := proc (n, x, M) option operator, arrow; local m;
    sum (a [n, m] * N (x, m) , m=1..M);
end proc;
```

$$N := (x, m) \rightarrow x^m (1-x)$$

$$\phi := (n, x, M) \rightarrow \text{local } m; \sum_{m=1}^M a_{n,m} N(x, m)$$

Процедура, формирующая уравнения метода взвешенных невязок и решающая полученную систему уравнений:

```
> WDM := proc (M, L) local n, m, eq, res, eqns, var, psi, R;
    psi [0] := 0;
    eqns := {}: var := {}:
    for n from 1 to L do
        psi [n] := phi (n, x, M);
        R := diff ((1+0.1*psi [n-1]) * diff (psi [n], x) , x) + 10*x;
        for m from 1 to M do
            eq [n, m] := int (R * N (x, m) , x=0..1) = 0;
        end do;
        eqns := eqns union {eq [n, m]}: var := var union {a [n, m]}:
    end do;
```

```

res:=fsolve(eqns,var);assign(res);
print(res);
end do;
end proc:

```

Данная задача допускает точное решение. С целью сравнения приближенного и точного решений, построим это точное решение с помощью Maple. Для этого зададим уравнение и граничные условия

```

> EQ:=diff((1+0.1*y(x))*diff(y(x),x),x)=-10*x;
bound:=y(0)=0,y(1)=0;

```

$$EQ := .1 \left(\frac{\partial}{\partial x} y(x) \right)^2 + (1 + .1 y(x)) \left(\frac{\partial^2}{\partial x^2} y(x) \right) = -10 x$$

$$bound := y(0) = 0, y(1) = 0$$

Получим решение и определим разрешающую функцию

```

> RES:=dsolve({EQ,bound},y(x));
assign(RES);y(x):
y:=unapply(y(x),x);

```

$$RES := y(x) = -10 + \frac{1}{3} \sqrt{900 - 300x^3 + 300x}$$

$$y := x \rightarrow -10 + \frac{1}{3} \sqrt{900 - 300x^3 + 300x}$$

Проверим полученное решение

```

> simplify(y(0));simplify(y(1));
0
0
> simplify(diff((1+0.1*y(x))*diff(y(x),x),x));
-10.x

```

Перейдем теперь к аппроксимации разрешающей функции методом взвешенных невязок по разработанной процедуре. Получим двухэлементную аппроксимацию при четырех итерациях:

```

> M:=2;L:=4;a:=array(1..L,1..M):WDM(M,L);
M:=2
L:=4

```

$$\{a_{1,2} = 1.666666666, a_{1,1} = 1.666666667\}$$


$$\{a_{2,1} = 1.633539066, a_{2,2} = 1.609101773, a_{1,2} = 1.666666667, a_{1,1} = 1.666666666\}$$




$\{a_{3,1} = 1.634308216, a_{3,2} = 1.610573404, a_{2,1} = 1.633539066, a_{2,2} = 1.609101773, a_{1,2} = 1.666666667,$
 $a_{1,1} = 1.666666666\}$

$\{a_{4,1} = 1.634289675, a_{3,1} = 1.634308216, a_{3,2} = 1.610573404, a_{2,1} = 1.633539066, a_{2,2} = 1.609101773,$
 $a_{1,2} = 1.666666667, a_{1,1} = 1.666666666, a_{4,2} = 1.610537192\}$

Сравнение точного решения и приближенного, полученного двухэлементной аппроксимацией

```
> pt:=plot(y(x),x=0..1,color=black,
  legend=`точное решение`):
p1:=plot(phi(1,x,M),x=0..1,color=black,style=point,
  legend=`первая итерация`):
p2:=plot(phi(2,x,M),x=0..1,color=black,
  style=point,symbol=circle,
  legend=`вторая итерация`):
p3:=plot(phi(3,x,M),x=0..1,color=black,
  style=point,symbol=box,
  legend=`третья итерация`):
p4:=plot(phi(4,x,M),x=0..1,color=black,
  style=point,symbol=cross,
  legend=`четвертая итерация`):
plots[display]({pt,p1});
plots[display]({pt,p2});
plots[display]({pt,p3});
plots[display]({pt,p4});
```





В заключение сделаем несколько замечаний. В этой главе для решения обыкновенных дифференциальных уравнений и уравнений с частными производными мы рассмотрели процесс аппроксимации базисными функциями, который является альтернативой для численных методов, таких как, например, метод конечных разностей. Все рассмотренные здесь методы можно отнести к численно-аналитическим методам поскольку, в конечном счете, решение получается в виде аналитической формулы. Полезность рассмотренных подходов очевидна. Вместе с тем очевидны и недостатки аппроксимации базисными функциями. Для двух- и, тем более, трехмерных областей использование базисных функций ограничивает нас рассмотрением прямоугольников, параллелепипедов и других областей простой формы, если мы хотим точно учесть граничные условия. Кроме того, матрица системы алгебраических уравнений, получающаяся при использовании методов взвешенных невязок, с увеличением числа удерживаемых для аппроксимации базисных функций в ряде случаев может стать плохо обусловленной. Правда, последняя проблема легко решается в Maple простым переопределением константы `Digits`.

Упражнения

1. Распределение изгибающего момента M в балке под действием распределенной поперечной нагрузки $q(x)$ на единицу длины удовлетворяет уравнению $\frac{d^2 M}{dx^2} = q(x)$. Балка единичной длины свободно оперта (то есть $M = 0$) на обоих концах и несет нагрузку $q(x) = \sin \pi x$ на единицу длины. Найти распределение изгибающего момента, используя методы поточечной коллокации, Галеркина и наименьших квадратов и подходящую систему базисных функций. Сравнить ответы с точным решением.
2. Некоторая задача одномерной стационарной теплопроводности с распределенным источником тепла описывается уравнением $\frac{d^2 \varphi}{dx^2} + \varphi + 1 = 0$ с краевыми условиями $\varphi = 0$ при $x = 0$ и $\frac{d\varphi}{dx} = -\varphi$ при $x = 1$. Найти приближенное решение методом Галеркина и исследовать скорость сходимости метода, сравнив полученные результаты с точным решением.
3. Уравнение $EI \frac{d^4 w}{dx^4} + kw = q(x)$ описывает отклонение балки, покоящейся на упругом основании жесткостью k . Здесь EI — постоянная жесткость балки на изгиб, а $q(x)$ — поперечная нагрузка на единицу ее длины. Пусть балка имеет единичную длину и закреплена в обоих концах так, что $\frac{dw}{dx} = w = 0$ при $x = 0$ и при $x = 1$. Найти отклонение нагруженной балки на упругом основании методами поточечной коллокации и Галеркина. Сравнить ответы с точным решением. При численных расчетах принять $\frac{q}{EI} = \frac{k}{EI} = 1$.
4. В некоторой двумерной задаче стационарной теплопроводности для квадрата со стороной длины 2 температура на сторонах $x = \pm 1$ изменяется как $1 - y^2$, а на сторонах $y = \pm 1$ изменяется как $1 - x^2$. Используя метод Галеркина, найти распределение температуры на квадрате.
5. Однородная тонкая упругая пластина Ω с постоянной жесткостью D на изгиб свободно оперта вдоль сторон Γ и находится под действием единичной равномерно распределенной поперечной нагрузки на единицу площади. Малое поперечное отклонение такой пластины описывается дифференциальным уравнением $\frac{\partial^4 w}{\partial x^4} + 2 \frac{\partial^4 w}{\partial x^2 \partial y^2} + \frac{\partial^4 w}{\partial y^4} = \frac{1}{D}$ в Ω и краевыми условиями $w = \frac{\partial^2 w}{\partial n^2} = 0$ на Γ . Найти отклонение для прямоугольной пластины $|x| \leq 3$, $|y| \leq 2$, используя методы поточечной коллокации и Галеркина.

6. Решить уравнение $\frac{d^2\varphi}{dx^2} + \varphi = 0$ с краевыми условиями $\varphi = 1$ при $x = 0$ и $\varphi = 0$ при $x = 1$. Применить метод взвешенных невязок, используя сначала аппроксимацию, удовлетворяющую заданным краевым условиям, а затем аппроксимацию, для которой краевые условия не удовлетворяются автоматически. Сравнить результаты с поведением точного решения.
7. Методом взвешенных невязок найти распределение изгибающего момента в балке из задачи 1, используя аппроксимацию, удовлетворяющую краевому условию при $x = 0$, но не удовлетворяющую автоматически краевому условию при $x = 1$. Использовать граничные весовые функции $\bar{W}_I = \alpha N_I|_{\Gamma}$, где α — постоянная, и сравнить результаты, найденные при $\alpha = \pm 0,1$; $\alpha = \pm 1$; $\alpha = \pm 10$; $\alpha = \pm 100$, с результатами из задачи 1, полученными с помощью аппроксимации, автоматически удовлетворяющей обоим краевым условиям.
8. В задаче 4 найти распределение температуры, используя аппроксимацию, удовлетворяющую краевым условиям только на сторонах $x = \pm 1$. Показать сходимость аппроксимации к краевому условию на сторонах $y = \pm 1$.
9. Найти приближенное решение задачи 5, которое сначала удовлетворяет только граничному условию нулевого смещения на краях пластины. Показать улучшение выполнения другого граничного условия на краях пластины с увеличением числа используемых в аппроксимации элементов.
10. Уравнение, описывающее изменение температуры T в вязкой жидкости, текущей между двумя параллельными пластинами $y = 0$ и $y = 2H$, имеет вид

$$\frac{d^2T}{dy^2} = -\frac{4U^2\mu}{H^4k}(H-y)^2,$$

где μ , k и U — коэффициент вязкости, коэффициент теплопроводности и максимальная скорость жидкости соответственно. Решить задачу методом Галеркина при условиях, что на одной пластине поддерживается температура $T = 0$, а через другую теплоотдача отсутствует $\frac{dT}{dy} = 0$. Сравнить точное и приближенное решения. В числовых расчетах принять $\mu = 0,1$, $k = 0,08$, $H = 3,0$ и $U = 3,0$.

11. Рассмотреть одномерную установившуюся теплопередачу в стержне длиной 10 см и диаметром 1 см. На одном конце стержня поддерживается температура 50°C , тогда как тепловая энергия поступает через другой конец со скоростью 200 Вт/см^2 . Пусть $k = 75 \text{ Вт/см}^0\text{C}$ и тепло генерируется в стержне со скоростью $150 T \text{ Вт/см}^2$ на единицу длины, где T — температура. Вычислить распределение температуры. Решить задачу методом Галеркина. Показать сходимость приближенного решения к точному при возрастании числа элементов, используемых в аппроксимации.
12. В задаче 3 выяснить, какие условия являются естественными и решить эту задачу методом Галеркина для случая, когда балка свободно оперта на концах, используя аппроксимацию, не удовлетворяющую автоматически этим естественным краевым

вым условиям. Краевые условия, соответствующие свободно опертой балке, имеют вид $\frac{d^2w}{dx^2} = w = 0$ на обоих концах.

13. Вывести уравнения метода взвешенных невязок для задачи двумерной теплопроводности в области Ω с краевыми условиями $\varphi = \bar{\varphi}$ на Γ_φ и $k \frac{\partial \varphi}{\partial n} = -\bar{q} + h(\varphi - \varphi_0)$ на Γ_q . Здесь φ — температура в произвольной точке, k — коэффициент теплопроводности среды, а $\bar{\varphi}$, \bar{q} , h и φ_0 — заданные функции координат.
14. Решить задачу стационарной теплопроводности в материале, занимающем квадрат $-1 \leq x \leq 1$, $-1 \leq y \leq 1$, если на сторонах $y = \pm 1$ поддерживается температура 100°C , тогда как на сторонах $x = \pm 1$ задано условие $\frac{d\varphi}{dn} = -1 - \varphi$.
15. Показать, что в задаче 5 об отклонении нагруженной тонкой упругой пластины краевое условие $\frac{\partial^2 w}{\partial n^2} = 0$ на Γ является естественным краевым условием. Решить эту задачу и сравнить ответы с полученными в задаче 9.
16. Решить уравнение $e^{-\varphi} \frac{d^2 \varphi}{dx^2} = 1$ с краевыми условиями $\varphi = 0$ при $x = 0$ и $x = 1$, используя методы поточечной коллокации и Галеркина. Это уравнение типично для задач о распространении тепла в химически реагирующих материалах, для которых Q пропорционально e^φ , где φ — температура.

Математическая теория поля

Часть пространства или все пространство, которому присущи некоторые физические свойства, принято называть *физическим полем*. Поле может быть *скалярным* или *векторным* в зависимости от характера исследуемой величины¹. Например, при рассмотрении потока газа приходится исследовать температурное поле (в каждой точке температура имеет определенное значение), поле плотностей, поле скоростей и т. п. Поля температур или плотностей являются скалярными, а поля скоростей или сил — векторными. Изучение физических полей мы начнем с рассмотрения скалярного поля. Большинство задач и примеров, рассматриваемых в этой главе, взяты из книги В.И. Семяннистого [9].

Скалярное поле

Если каждой точке P области Ω поставлено в соответствие вполне определенное значение $u(P)$ некоторой физической величины, то говорят, что в области Ω определено поле этой величины. Поле называется *скалярным*, если $u(P)$ — скалярная величина, то есть если ее значения в определенной системе единиц суть числа.

Если область Ω трехмерная, то понятие скалярного поля совпадает, в сущности, с понятием функции трех переменных $u(P) = u(x, y, z)$. Эта функция, независимо от ее физического смысла, называется *потенциалом поля*.

Если потенциал поля является только функцией точки $u(P) = u(x, y, z)$ и не зависит от времени t , то скалярное поле называется *стационарным*. В общем случае, когда $u = u(P, t) = u(x, y, z, t)$, поле называется *нестационарным*.

Скалярное поле $u(P)$ называется *плоским*, если значение $u(P)$ постоянно вдоль любого перпендикуляра, проведенного к некоторой фиксированной плоскости.

Поверхность, на которой значение поля остается постоянным, называется *поверхностью уровня* или *эквипотенциальной поверхностью*. Поверхность уровня определяется уравнением

$$u(P) = C, \quad (9.1)$$

где $C = const$.

Примерами эквипотенциальных поверхностей являются в поле температур изотермические поверхности, в поле атмосферного давления изобарические поверхности, наконец, изопотенциальные поверхности электрического поля потенциалов.

Семейство поверхностей уровня наглядно характеризует некоторые свойства изучаемого скалярного поля. Так, места сближения поверхностей указывают на быстрое изменение потенциала u в поперечном направлении.

¹ В физике встречаются поля и более сложной природы, например, *тензорные поля*.

Методика изучения скалярного поля $u = u(x, y, z)$ с помощью поверхностей уровня заключается в следующем. В области изменения функции $u = u(x, y, z)$ выбирается произвольное значение $u = C$. Затем задается положительное число $\Delta C > 0$, такое, чтобы значения $C - n\Delta C, C - (n-1)\Delta C, \dots, C - \Delta C, C, C + \Delta C, \dots, C + n\Delta C$ не выходили из области изменения функции $u = u(x, y, z)$. После этого в системе координат x, y, z строятся поверхности равного уровня

$$\begin{aligned}
 u(x, y, z) &= C - n\Delta C, \\
 &\dots\dots\dots \\
 u(x, y, z) &= C - \Delta C, \\
 u(x, y, z) &= C, \\
 u(x, y, z) &= C + \Delta C, \\
 &\dots\dots\dots \\
 u(x, y, z) &= C + n\Delta C.
 \end{aligned}$$

Выделяются области сближения построенных поверхностей и области, где они значительно удалены друг от друга. Области первого типа характеризуют более быстрое изменение исследуемого поля в поперечном по отношению к поверхности уровня направлении, чем области второго типа.

Метод поверхностей уровня, как мы видим, не очень эффективен, так как построение самих поверхностей в системе координат x, y, z является весьма трудоемкой задачей. Но во многих практических задачах, особенно при исследовании плоских полей $u = u(x, y)$, этот метод дает хороший результат, и к нему часто прибегают.

Примеры

Задача 9.1. Найти линии уровня плоского поля $u = xy$.


Решение. Линии уровня определяются уравнением $xy = C$ и представляют собой равносторонние гиперболы. При $C = 0$ линией уровня является совокупность координатных осей Ox и Oy .

Воспользуемся Maple для иллюстрации. Для этого подключим пакет `plots` и отобразим линии уровня с помощью команды `contourplot`

```

> with(plots):
> contourplot(x*y, x=-3..3, y=-3..3, filled=true, grid=[50, 50],
  contours=[-2, -1, -3/4, -1/4, -1/2, 1/4, 1/2, 3/4, 1, 2]);

```



Команда `contourplot` отображает линии равного уровня и может использоваться в следующих форматах

```
contourplot(expr, x=a..b, y=c..d);
contourplot(f, a..b, c..d);
contourplot([exprf, exprg, exprh], s=a..b, t=c..d);
contourplot([f, g, h], a..b, c..d);
```

В первых двух форматах `expr`, `f` — выражения (функции), зависящие от x и y ; второй и третий аргументы задают диапазон изменения переменных x и y . Вторые два формата используются в случае параметрического задания поверхности.

Опция `filled=true` обеспечивает функциональную автоматическую окраску изображаемой области. Порой это придает большую выразительность, чем при построении только линий равного уровня.

Опция `contours=[-2, -1, -3/4, -1/4, -1/2, 1/4, 1/2, 3/4, 1, 2]` указывает те значения константы C , при которых требуется построить линии равного уровня.

Задача 9.2. Найти поверхности уровня скалярного поля $u = \operatorname{arctg} \left(\frac{\sqrt{x^2 + y^2}}{z} \right)$.

Решение. Поверхности уровня определяются уравнением (9.1). Отсюда находим

$$\frac{x^2 + y^2}{z^2} = a^2,$$

где $a = \operatorname{tg} C$, $-\frac{\pi}{2} < C < \frac{\pi}{2}$.

Таким образом, мы видим, что поверхностями уровня являются круговые конусы

$$x^2 + y^2 - a^2 z^2 = 0$$

с осью Oz .

Задача 9.3. Найти поверхности уровня скалярного поля $u = \frac{1}{3x + 2y - 7z + 5}$.

Решение. Скалярное поле определено для всех точек пространства, кроме точек, расположенных на плоскости $3x + 2y - 7z + 5 = 0$. Поверхности уровня определяются уравнениями

$$\frac{1}{3x + 2y - 7z + 5} = C_1 \Rightarrow 3x + 2y - 7z + 5 = C_2, C_2 = \frac{1}{C_1},$$

описывающими семейство параллельных плоскостей

$$3x + 2y - 7z + C = 0, C = 5 - C_2.$$

Задача 9.4. Найти поверхности уровня скалярного поля $u = x^2 + z^2 - y^2$.

Решение. Уравнение семейства поверхностей уровня имеет вид $x^2 + z^2 - y^2 = C$.


При $C = 0$ уравнение $x^2 + z^2 - y^2 = 0$ определяет конус. При $C > 0$ уравнение $x^2 + z^2 - y^2 = C$ определяет семейство однополостных гиперболоидов. При $C < 0$ уравнение $x^2 + z^2 - y^2 = C$ определяет семейство двуполостных гиперболоидов.

Задача 9.5. Горизонтали возвышенности определяются уравнением $h = 20 - \frac{x^2}{4} - y^2$.

Построить горизонтали, соответствующие отметкам $h = 20, 19, 18, 16$ и 11 метров.


Решение. Воспользуемся командой `contourplot`

```
> plots[contourplot](20-x^2/4-y^2,  
x=-7..7,y=-4..4,filled=true,grid=[25,25],  
contours=[11,16,18,19,20]);
```



или аналогичной командой `contourplot3d`, которая строит изображение в пространстве

```
> plots[contourplot3d](20-x^2/4-y^2,
x=-7..7,y=-4..4,filled=true,grid=[25,25],
contours=[11,16,18,19,20],axes=normal,view=8..20);
```



Производная скалярного поля по направлению

Пусть в области, где задано скалярное поле $u(P)$, взята точка P . Рассмотрим произвольный луч l , выходящий из точки P .

Производной поля $u(P)$ в точке P в направлении l называется предел

$$\frac{\partial u}{\partial l} = \lim_{Q \rightarrow P} \frac{u(Q) - u(P)}{|QP|}$$

при условии, что переменная точка Q стремится к точке P вдоль луча l .

Для вычисления производной по направлению в декартовых координатах служит формула

$$\frac{\partial u}{\partial l} = \frac{\partial u}{\partial x} \cos \alpha + \frac{\partial u}{\partial y} \cos \beta + \frac{\partial u}{\partial z} \cos \gamma, \quad (9.2)$$

где α, β, γ — углы, которые луч l образует с осями координат.

Если направление l задано с помощью некоторого вектора $\vec{l} = l_x \vec{i} + l_y \vec{j} + l_z \vec{k}$, где $\vec{i}, \vec{j}, \vec{k}$ — орты прямоугольной декартовой системы координат, то, как известно,

$$\cos \alpha = \frac{l_x}{|\vec{l}|} = \frac{l_x}{\sqrt{l_x^2 + l_y^2 + l_z^2}},$$

$$\cos \beta = \frac{l_y}{|\vec{l}|} = \frac{l_y}{\sqrt{l_x^2 + l_y^2 + l_z^2}},$$

$$\cos \gamma = \frac{l_z}{|\vec{l}|} = \frac{l_z}{\sqrt{l_x^2 + l_y^2 + l_z^2}}. \quad (9.3)$$

В частности, если вектор \vec{l} — единичный, то $\cos \alpha = l_x$, $\cos \beta = l_y$, $\cos \gamma = l_z$.

Производная поля $u(P)$ в точке P по направлению произвольной линии L совпадает с производной по направлению касательной к этой линии в точке P . Касательный вектор \vec{l} к линии L определяется из уравнений линии $x = x(t)$, $y = y(t)$, $z = z(t)$ по формулам $l_x = x'(t)$, $l_y = y'(t)$, $l_z = z'(t)$.

Если в качестве параметра t берется длина дуги s , то вектор \vec{l} — единичный, и предыдущие формулы принимают вид $\cos \alpha = x'(s)$, $\cos \beta = y'(s)$, $\cos \gamma = z'(s)$.

Примеры

Задача 9.6. Найти производную поля $u = y^2z - 2xyz + z^2$ в точке $P_0(3, 1, 1)$ по направлению вектора \vec{l} , если этот вектор образует с координатными осями Ox и Oy соответственно углы $\alpha = \frac{\pi}{3}$, $\beta = \frac{\pi}{4}$, а угол, образованный вектором \vec{l} с осью Oz , равен γ , причем $\gamma < \frac{\pi}{2}$.

Решение. Производную по направлению вектора \vec{l} следует вычислять по формуле (9.2), в которой, очевидно, $\cos \gamma = \sqrt{1 - \cos^2 \alpha - \cos^2 \beta}$. Выполним все расчеты в Maple. Вычислить производную по заданному направлению можно с помощью команды `DirectionalDiff`, которая находится в пакете `VectorCalculus`. Поэтому мы сразу подключим этот пакет командой

```
> with(VectorCalculus):
Warning, the assigned names <,> and <|> now have a global binding
Warning, these protected names have been redefined and unprotected: *, +, ., Vector, diff, int, limit, series
```

При подключении пакета Maple предупреждает нас, что теперь некоторые назначенные обозначения глобально связаны и некоторые защищенные обозначения и имена будут переопределены. Это необходимо учитывать в своей дальнейшей работе.

Определим заданное поле

```
> u := y^2*z - 2*x*y*z + z^2;
u := y^2 z - 2 x y z + z^2
```

Определим направление, вдоль которого надо вычислить производную, в виде вектора

```
> L:=<cos(Pi/3),cos(Pi/4),sqrt(1-cos(Pi/3)^2-cos(Pi/4)^2)>;
```

$$L := \frac{1}{2} \mathbf{e}_x + \frac{\sqrt{2}}{2} \mathbf{e}_y + \frac{1}{2} \mathbf{e}_z$$

Вычисляем теперь производную по заданному направлению с помощью команды DirectionalDiff

```
> DirectionalDiff( u, L, [x,y,z] );
```

$$-y z + \frac{(2 y z - 2 x z) \sqrt{2}}{2} + \frac{y^2}{2} - x y + z$$

Первым аргументом команды DirectionalDiff выступает алгебраическое выражение, задающее поле, второй аргумент определяет вектор, в направлении которого ищется производная, третий аргумент задает список координат (переменных).

Для того чтобы вычислить значение этой производной в заданной точке, воспользуемся командой подстановки subs

```
> subs(x=3,y=1,z=1,%);
```

$$-2\sqrt{2} - \frac{5}{2}$$

Итак, решение нашей задачи: $\left. \frac{\partial u}{\partial l} \right|_{\substack{x=3 \\ y=1 \\ z=1}} = -\frac{5 + 4\sqrt{2}}{2}$.

Задача 9.7. Найти производную скалярного поля $u = xyz$ в точке $P_0(1,1,1)$ по направлению вектора $\vec{l} = \vec{i} - 2\vec{j} + 2\vec{k}$.

Решение. В системе Maple задача решается так:

```
> u:=x*y*z;
```

$$u := x y z$$

```
> L:=<1,-2,2>;
```

$$L := \mathbf{e}_x - 2 \mathbf{e}_y + 2 \mathbf{e}_z$$

```
> DirectionalDiff( u, L, [x,y,z] );
```

$$\frac{1}{3} y z - \frac{2}{3} x z + \frac{2}{3} x y$$

```
> subs(x=3,y=1,z=1,%);
```

$$\frac{1}{3}$$

Мы предполагаем, что пакет `VectorCalculus` уже был подключен. Если это не так, то можно не подключая всего пакета использовать команду

```
> VectorCalculus[DirectionalDiff](u,L,[x,y,z]);
```

Итак, решение нашей задачи: $\left. \frac{\partial u}{\partial l} \right|_{\substack{x=1 \\ y=1 \\ z=1}} = \frac{1}{3}$.

Задача 9.8. Найти производную поля $u(P) = 4x^3 - 3y^3$ в точке $N(1,1)$ по направлению вектора \vec{l} , идущему от этой точки к точке $M(4,5)$.

Решение. Воспользуемся формулой (9.2). Для этого вычислим частные производные поля по x и по y в точке $N(1,1)$. В системе Maple задача решается, например, так. Определим поле $u(P) = u(x, y)$ в виде функции

```
> u := (x, y) -> 4*x^3 - 3*y^3;
```

$$u := (x, y) \rightarrow 4x^3 - 3y^3$$

Теперь находим частные производные этого поля по x и по y в точке $N(1,1)$. Проще всего это сделать с помощью оператора `D` так

```
> dx := D[1](u)(1,1); dy := D[2](u)(1,1);
```

$$dx := 12$$

$$dy := -9$$

Здесь мы значения частных производных поля по x и по y в точке $N(1,1)$ обозначили соответственно через `dx` и `dy`.

Теперь определяем направление, в котором нам надо вычислить производную. Задаем точки M и N , например, так

```
> M := [4, 5]; N := [1, 1];
```

$$M := [4, 5]$$

$$N := [1, 1]$$

Определяем координаты отрезка L , соединяющего эти точки

```
> l[1] := M[1] - N[1]; l[2] := M[2] - N[2];
```

$$l_1 := 3$$

$$l_2 := 4$$

и вычисляем длину отрезка L

```
> L := <l[1], l[2]>; Length := sqrt(l[1]^2 + l[2]^2);
```

$$L := \begin{bmatrix} 3 \\ 4 \end{bmatrix}$$

$$\text{Length} := 5$$

Теперь вычисляем направляющие косинусы отрезка L по формулам (9.3)

$$> m := L[1] / \text{Length}; n := L[2] / \text{Length};$$

$$m := \frac{3}{5}$$

$$n := \frac{4}{5}$$

И, наконец, вычисляем производную по направлению L в точке $N(1,1)$

$$> \text{'Diff}(u, L) = dx * m + dy * n;$$

$$\frac{\partial}{\partial L} u = 0$$

Итак, решение нашей задачи: $\left. \frac{\partial u}{\partial L} \right|_{\substack{x=1 \\ y=1}} = 0$. Равенство нулю производной означает, что

вектор \vec{l} направлен по касательной к линии уровня.

Эту же задачу проще решить с помощью пакета `VectorCalculus` так:

$$> \text{with}(\text{VectorCalculus}) :$$

$$> U := u(x, y);$$

$$U := 4x^3 - 3y^3$$

$$> L := \langle L[1], L[2] \rangle ;$$

$$L := 3 \mathbf{e}_x + 4 \mathbf{e}_y$$

$$> L_ort := L / \text{Length};$$

$$L_ort := \frac{3}{5} \mathbf{e}_x + \frac{4}{5} \mathbf{e}_y$$

$$> \text{DirectionalDiff}(U, L, [x, y]);$$

$$\frac{36x^2}{5} - \frac{36y^2}{5}$$

$$> \text{subs}(x=1, y=1, \%);$$

0

Здесь мы попутно определили орт направления L (переменная `L_ort`), хотя для решения нашей задачи с помощью пакета `VectorCalculus` не требуется вычисление орта направления — необходимо только задать выражение поля в зависимости от x и y (переменная `U`), вектор, вдоль которого ищется производная (переменная `L`) и список координат (переменных). Команда `DirectionalDiff` вычисляет производную по направлению в зависимости от координат x и y , то есть в любой точке.

Чтобы вычислить производную по направлению в конкретной заданной точке, надо просто подставить в общее выражение этой производной значения координат заданной точки. Это делается с помощью команды подстановки `subs`.

Задача 9.9. Установить характер изменения скалярного поля $u(P) = xyz$ в точке $N(5, 1, -8)$ в направлении, идущем от этой точки к точке $M(9, 4, 4)$.

Решение. В Maple задача решается так:

```
> restart;
> u:=x*y*z;
                                     u := x y z
> M:=[9,4,4];N:=[5,1,-8];
                                     M := [9, 4, 4]
                                     N := [5, 1, -8]
> l[1]:=M[1]-N[1];l[2]:=M[2]-N[2];l[3]:=M[3]-N[3];
                                     l1 := 4
                                     l2 := 3
                                     l3 := 12
> L:=<l[1],l[2],l[3]>;
                                     L :=  $\begin{bmatrix} 4 \\ 3 \\ 12 \end{bmatrix}$ 
> VectorCalculus[DirectionalDiff](u,L,[x,y,z]);
                                      $\frac{4}{13}yz + \frac{3}{13}xz + \frac{12}{13}xy$ 
> subs(x=5,y=1,z=-8,%);
                                      $\frac{-92}{13}$ 
```

Таким образом, мы получили $\left. \frac{\partial u}{\partial l} \right|_{\substack{x=5 \\ y=1 \\ z=-8}} = -\frac{92}{13}$. Видим, что производная в направлении

вектора \overline{NM} отрицательна. Значит, данное скалярное поле убывает в этом направлении (в точке N) со скоростью, равной

$$v = \left| \frac{\partial u}{\partial l} \right| = \frac{92}{13}.$$

Задача 9.10. Найти производную скалярного поля $u(P) = x^2 + yz$ по направлению винтовой линии $x = a \cos t$, $y = a \sin t$, $z = bt$ в точке N , где $t = \frac{\pi}{2}$.

Решение. Решаем задачу Maple. Подключаем пакет VectorCalculus и задаем параметрические уравнения винтовой линии:

```
> restart;
> with(VectorCalculus):
Warning, the assigned names <, > and <|> now have a global binding
Warning, these protected names have been redefined and unprotected: *, +, ., Vector, diff, int, limit, series
> x:=a*cos(t);y:=a*sin(t);z:=b*t;
      x := a cos(t)
      y := a sin(t)
      z := b t
```

Очевидно, данная винтовая линия расположена на цилиндре $x^2 + y^2 = a^2$. Точка N имеет координаты

```
> N:=[x(Pi/2),y(Pi/2),z(Pi/2)];
      N := [ 0, a,  $\frac{b \pi}{2}$  ]
```

То есть, эта точка является точкой пересечения винтовой линии с плоскостью yOz .

Определяем радиус-вектор

```
> r:=<x,y,z>;
      r := a cos(t)  $\mathbf{e}_x$  + a sin(t)  $\mathbf{e}_y$  + b t  $\mathbf{e}_z$ 
```

и направление, вдоль которого требуется найти производную заданного скалярного поля

```
> L:=subs(t=Pi/2,diff(r,t));
      L := -a  $\mathbf{e}_x$  + b  $\mathbf{e}_z$ 
```

Теперь вычисляем производную вдоль этого направления

```
> x:='x':y:='y':z:='z':
value(DirectionalDiff(x^2+y*z,L,[x,y,z]));
       $\frac{-2 x a + y b}{\sqrt{a^2 + b^2}}$ 
```

и значение производной в точке N

> subs (x=N[1], y=N[2], z=N[3], %) ;

$$\frac{a b}{\sqrt{a^2 + b^2}}$$

Итак, окончательно получаем

$$\frac{\partial u}{\partial l} = \frac{ab}{\sqrt{a^2 + b^2}} .$$

Градиент скалярного поля

Одним из основных понятий теории поля является понятие *градиента скалярного поля*. Оказывается, градиент скалярного поля и производная по направлению теснейшим образом связаны друг с другом. Но связь эта особая.

Действительно, правую часть равенства (9.2) можно рассматривать как скалярное произведение двух векторов. Один из этих векторов имеет проекции $\frac{\partial u}{\partial x}$, $\frac{\partial u}{\partial y}$, $\frac{\partial u}{\partial z}$.

Проекциями второго вектора будут $\cos \alpha$, $\cos \beta$, $\cos \gamma$, то есть вектором-множителем будет направляющий вектор прямой l единичной длины: $\vec{l}_0 = \vec{i} \cos \alpha + \vec{j} \cos \beta + \vec{k} \cos \gamma$, ($|\vec{l}_0| = 1$).

Вектор

$$\text{grad } u = \frac{\partial u}{\partial x} \vec{i} + \frac{\partial u}{\partial y} \vec{j} + \frac{\partial u}{\partial z} \vec{k} , \quad (9.4)$$

участвующий в образовании производной по направлению, называется *градиентом поля*.

Заметим, что, для того чтобы скалярное поле имело в каждой точке градиент с определенным конечным значением, необходимо, чтобы функция $u(x, y, z)$ удовлетворяла условиям однозначности, непрерывности, отсутствия особых точек поверхностей уровня.

Отмеченная выше связь градиента с производной по направлению выражается, очевидно, формулой

$$\frac{\partial u}{\partial l} = \text{grad } u \cdot \vec{l}_0 . \quad (9.5)$$

Градиент скалярного поля $u(P)$ в точке P направлен по нормали к поверхности уровня $u(x, y, z) = C$, проходящей через точку P .

Из формулы (9.5) видно, что функция $u(x, y, z)$ наиболее сильно возрастает в направлении градиента. Другими словами, учитывая, что $\frac{\partial u}{\partial l}$ характеризует скорость изменения потенциала поля, можно сказать, что $\text{grad } u$ есть вектор, в направлении которого скорость изменения потенциала наибольшая, причем длина вектора

$$|\text{grad } u| = \sqrt{\left(\frac{\partial u}{\partial x}\right)^2 + \left(\frac{\partial u}{\partial y}\right)^2 + \left(\frac{\partial u}{\partial z}\right)^2}$$

равна величине этой наибольшей скорости.

С помощью градиента можно найти единичный вектор нормали \vec{n} к поверхности уровня $u(x, y, z) = C$. Вектор \vec{n} определяется формулой

$$\vec{n} = \frac{\text{grad } u}{|\text{grad } u|}.$$

Градиент скалярной функции $f(r)$, $r = \sqrt{x^2 + y^2 + z^2}$ определяется следующей формулой

$$\text{grad } f(r) = \frac{\partial f}{\partial r} \vec{r}_0, \quad (9.6)$$

где $\vec{r}_0 = \frac{\vec{r}}{r}$ — единичный вектор в направлении $\vec{r} = x\vec{i} + y\vec{j} + z\vec{k}$.

Введя понятие градиента скалярного поля $u(x, y, z)$, мы определили его как вектор (9.4) с координатами $\frac{\partial u}{\partial x}$, $\frac{\partial u}{\partial y}$, $\frac{\partial u}{\partial z}$. Удобно в формуле (9.4), означающей переход от скалярного поля $u(x, y, z)$ к векторному полю $\text{grad } u$, операции дифференцирования рассматривать как операции умножения символов $\frac{\partial}{\partial x}$, $\frac{\partial}{\partial y}$, $\frac{\partial}{\partial z}$ на скалярную функцию $u(x, y, z)$. Тогда правая часть этой формулы может быть переписана в виде произведения символического (условного) вектора $\frac{\partial}{\partial x} \vec{i} + \frac{\partial}{\partial y} \vec{j} + \frac{\partial}{\partial z} \vec{k}$ на функцию $u(x, y, z)$, то есть

$$\text{grad } u = \left(\frac{\partial}{\partial x} \vec{i} + \frac{\partial}{\partial y} \vec{j} + \frac{\partial}{\partial z} \vec{k} \right) u.$$

Символический множитель-вектор, указывающий, какие операции должны быть проделаны над множителем u , называется *оператором Гамильтона* или *набла-оператором*. Обозначается оператор набла символом ∇

$$\nabla = \frac{\partial}{\partial x} \vec{i} + \frac{\partial}{\partial y} \vec{j} + \frac{\partial}{\partial z} \vec{k}.$$

Таким образом,

$$\text{grad } u = \nabla u.$$

Примеры

Задача 9.11. Дано скалярное поле $\ln\left(\frac{1}{\rho}\right)$, где $\rho = \sqrt{(x-a)^2 + (y-b)^2 + (z-c)^2}$. В каких точках пространства $Oxyz$ выполняется равенство $|\text{grad } u| = 1$?

Решение. Удобнее всего решать задачу в сферических координатах. Если поместить начало координат в точку $M(a, b, c)$, то ρ будет выражать расстояние от этой точки до произвольной точки пространства.

Решаем задачу в Maple. Подключаем пакет VectorCalculus:

```
> restart;
> with(VectorCalculus):
Warning, the assigned names <, > and <|> now have a global
binding
Warning, these protected names have been redefined and
unprotected: *, +, ., Vector, diff, int, limit, series
```

Задаем систему координат и определяем скалярное поле

```
> SetCoordinates('spherical'[rho, theta, phi]);
```

spherical _{ρ, θ, ϕ}

```
> u := ln(1/rho);
```

$$u := \ln\left(\frac{1}{\rho}\right)$$

Здесь командой SetCoordinates мы определили свойства координатной системы: теперь переменные rho, theta, phi представляют собой сферические координаты. Множество координатных систем, поддерживаемых Maple, можно посмотреть с помощью справки — набрать в строке ?Coordinates и нажать клавишу ввода. В учебной практике используются три наиболее употребительных системы координат: сферическая система координат; прямоугольная декартова система координат, которая устанавливается с помощью команды

```
> SetCoordinates('cartesian'[x, y, z]);
```

и цилиндрическая система координат, которая устанавливается с помощью команды

```
> SetCoordinates('cylindrical'[r, phi, z]);
```

Обозначения самих координат могут быть любые.

Вернемся к нашей задаче. Вычисляем градиент поля

> Gradient(u);

$$-\frac{1}{\rho} \bar{\mathbf{e}}_{\rho}$$

Таким образом, очевидно, $|\text{grad } u| = \frac{1}{\rho}$, и, следовательно, равенство $|\text{grad } u| = 1$ выполняется на сфере единичного радиуса с центром в точке $M(a, b, c)$, то есть на множестве точек $\rho = 1$.

Задача 9.12. Указать направление и скорость наибольшего возрастания скалярного поля $u = \sqrt{4 + x^2 + y^2}$ в точке $M(2, 1, 3)$.

Решение. Выбираем прямоугольную декартову систему координат и определяем поле

> SetCoordinates('cartesian'[x, y, z]);

*cartesian*_{x,y,z}

> u:=sqrt(4+x^2+y^2);

$$u := \sqrt{x^2 + 4 + y^2}$$

Вычисляем градиент этого поля

> Gradient(u);

$$\frac{x}{\sqrt{x^2 + 4 + y^2}} \bar{\mathbf{e}}_x + \frac{y}{\sqrt{x^2 + 4 + y^2}} \bar{\mathbf{e}}_y$$

В точке $M(2, 1, 3)$ это будет вектор

> gradM:=simplify(subs(x=2,y=1,%));

$$\text{grad}M := \frac{2}{3} \bar{\mathbf{e}}_x + \frac{1}{3} \bar{\mathbf{e}}_y$$

Вычисляем теперь модуль этого вектора

> mod_grad_M:=sqrt((2/3)^2+(1/3)^2);

$$\text{mod_grad_M} := \frac{\sqrt{5}}{3}$$

и находим его направление, вычислив направляющие косинусы по известным формулам

$$\cos \alpha = \frac{\text{Пр}_{Ox} \text{grad } u}{|\text{grad } u|}, \quad \cos \beta = \frac{\text{Пр}_{Oy} \text{grad } u}{|\text{grad } u|}.$$

> 'cos(alpha)'=2/3/mod_grad_M; 'cos(beta)'=1/3/mod_grad_M;

$$\cos(\alpha) = \frac{2\sqrt{5}}{5}$$

$$\cos(\beta) = \frac{\sqrt{5}}{5}$$

Чтобы не вводить координаты вектора с клавиатуры, всякий раз, когда это нужно, можно поступить и так — сначала определить орты прямоугольной декартовой системы координат:

```
> i:=VectorField(<1,0,0>);j:=VectorField(<0,1,0>);
```

$$i := \bar{\mathbf{e}}_x$$

$$j := \bar{\mathbf{e}}_y$$

затем выделить координаты градиента

```
> m:=i.gradM;n:=j.gradM;
```

$$m := \frac{2}{3}$$

$$n := \frac{1}{3}$$

Координаты градиента (и вообще, любого вектора) находим как скалярное произведение градиента на орты координатной системы. Скалярное произведение определяется просто точкой, поставленной между векторами (напомним, что после подключения пакета `VectorCalculus` некоторые общепринятые обозначения были переопределены, в частности, теперь точка означает скалярное произведение векторов). Отметим, что скалярное произведение векторов можно найти и с помощью команды `DotProduct` пакета `VectorCalculus`. Формат этой команды такой: `DotProduct(v1,v2)`. Здесь `v1,v2` — векторы или дифференциальные операторы.

Находим теперь модуль градиента

```
> mod_grad_M:=sqrt(m^2+n^2);
```

$$mod_grad_M := \frac{\sqrt{5}}{3}$$

и, наконец, вычисляем направляющие косинусы

```
> 'cos(alpha)'=m/mod_grad_M;'cos(beta)'=n/mod_grad_M;
```

$$\cos(\alpha) = \frac{2\sqrt{5}}{5}$$

$$\cos(\beta) = \frac{\sqrt{5}}{5}$$

На самом деле координаты вектора в Maple можно определить проще, — указав идентификатор вектора с индексом, определяющим номер координаты. Например, в нашем случае координаты вектора `gradM` можно выделить так

```
> m:=gradM[1];n:=gradM[2];
```

$$m := \frac{2}{3}$$

$$n := \frac{1}{3}$$

Задача 9.13. Пусть $u = xy - z^2$. Найти величину и направление $\text{grad} u$ в точке $M(-9, 12, 10)$. Чему равна производная $\frac{\partial u}{\partial l}$ в направлении биссектрисы координатного угла xOy ?

Решение. Выбираем прямоугольную декартову систему координат и определяем поле

```
> with(VectorCalculus):
```

```
Warning, the assigned names <,> and <|> now have a global binding
```

```
Warning, these protected names have been redefined and unprotected: *, +, ., Vector, diff, int, limit, series
```

```
> SetCoordinates('cartesian'[x,y,z]);
```

*cartesian*_{*x,y,z*}

```
> u:=x*y-z^2;
```

$$u := x y - z^2$$

Вычисляем градиент поля

```
> Gradient(u);
```

$$y \bar{\mathbf{e}}_x + x \bar{\mathbf{e}}_y - 2z \bar{\mathbf{e}}_z$$

Градиент в точке $M(-9, 12, 10)$

```
> gradM:=simplify(subs(x=-9,y=12,z=10,%));
```

$$\text{grad}M := 12 \bar{\mathbf{e}}_x - 9 \bar{\mathbf{e}}_y - 20 \bar{\mathbf{e}}_z$$

Определим орты системы координат, выделим координаты градиента в точке $M(-9, 12, 10)$ и найдем модуль этого градиента

```
> i:=VectorField(<1,0,0>);
```

```
j:=VectorField(<0,1,0>);
```

```
k:=VectorField(<0,0,1>);
```

$$i := \bar{\mathbf{e}}_x$$


```

j :=  $\bar{\mathbf{e}}_y$ 
k :=  $\bar{\mathbf{e}}_z$ 
> m:=i.gradM;n:=j.gradM;l:=k.gradM;
m := 12
n := -9
l := -20
> mod_grad_M:=sqrt(m^2+n^2+l^2);
mod_grad_M := 25

```

Направление градиента определяется направляющими косинусами

```

> 'cos(alpha)'=m/mod_grad_M;
'cos(beta)'=n/mod_grad_M;
'cos(gamma)'=l/mod_grad_M;
cos( $\alpha$ ) =  $\frac{12}{25}$ 
cos( $\beta$ ) =  $\frac{-9}{25}$ 
cos( $\gamma$ ) =  $\frac{-4}{5}$ 

```

или, что то же самое, вектором

```

> eM:=gradM/mod_grad_M;
eM :=  $\frac{12}{25} \bar{\mathbf{e}}_x - \frac{9}{25} \bar{\mathbf{e}}_y - \frac{4}{5} \bar{\mathbf{e}}_z$ 

```

Найдем теперь производную в направлении биссектрисы координатного угла xOy . Для этого определим направление биссектрисы и найдем производную в этом направлении как скалярное произведение градиента на вектор направления

```

> tau:=VectorField(<cos(Pi/4),cos(Pi/4),0>);
tau :=  $\frac{\sqrt{2}}{2} \bar{\mathbf{e}}_x + \frac{\sqrt{2}}{2} \bar{\mathbf{e}}_y$ 
> 'Diff(u,tau)'=gradM.tau;
 $\frac{\partial}{\partial \tau} u = \frac{3\sqrt{2}}{2}$ 

```

Задача 9.14. Найти $\text{grad}(\bar{c} \cdot \bar{r})$, где \bar{c} — постоянный вектор.

Решение. В системе Maple задача решается так:

```

> with(VectorCalculus):
Warning, the assigned names <,> and <|> now have a global
binding

Warning, these protected names have been redefined and
unprotected: *, +, .., Vector, diff, int, limit, series

> c:=VectorField(<A,B,C>);

$$c := A \bar{\mathbf{e}}_x + B \bar{\mathbf{e}}_y + C \bar{\mathbf{e}}_z$$


> r:=VectorField(<x,y,z>);

$$r := x \bar{\mathbf{e}}_x + y \bar{\mathbf{e}}_y + z \bar{\mathbf{e}}_z$$


> Gradient(c.r);

$$A \bar{\mathbf{e}}_x + B \bar{\mathbf{e}}_y + C \bar{\mathbf{e}}_z$$


```

Итак, мы получили $\text{grad}(\bar{c} \cdot \bar{r}) = \bar{c}$.

Задача 9.15. Найти $\text{grad}\left(\frac{1}{r}\right)$.

Решение. В системе Maple задача решается так:

```

> with(VectorCalculus):
Warning, the assigned names <,> and <|> now have a global
binding

Warning, these protected names have been redefined and
unprotected: *, +, .., Vector, diff, int, limit, series

> u:=1/sqrt(x^2+y^2+z^2);

$$u := \frac{1}{\sqrt{x^2 + y^2 + z^2}}$$


> Gradient(u, 'cartesian'[x,y,z]);

$$-\frac{x}{(x^2 + y^2 + z^2)^{(3/2)}} \bar{\mathbf{e}}_x - \frac{y}{(x^2 + y^2 + z^2)^{(3/2)}} \bar{\mathbf{e}}_y - \frac{z}{(x^2 + y^2 + z^2)^{(3/2)}} \bar{\mathbf{e}}_z$$


```

Мы видим, что если привести к общему знаменателю, то в числителе будем иметь вектор $\bar{r} = x\bar{i} + y\bar{j} + z\bar{k}$, а в знаменателе величину r^3 , где $r = \sqrt{x^2 + y^2 + z^2}$. Таким образом, $\text{grad}\left(\frac{1}{r}\right) = -\frac{\bar{r}}{r^3}$ или $\text{grad}\left(\frac{1}{r}\right) = -\frac{1}{r^2}\bar{r}_0$, где \bar{r}_0 — единичный вектор в направлении \bar{r} . Все эти преобразования можно выполнить в Maple с помощью следующей команды

```

> assume(rho>0):

```

```
simplify(MapToBasis( %, 'spherical'[rho,theta,phi] ));
```

$$-\frac{1}{\rho^2} \bar{\mathbf{e}}_\rho$$

Здесь мы воспользовались командой `MapToBasis`, которая переводит векторы и векторные поля из одних координатных систем в другие. Эта команда имеет следующий формат `MapToBasis(V, c)`. Здесь V — вектор, подлежащий преобразованию, c — указывает координатную систему, в которой должен быть представлен вектор.

Эту же задачу можно сразу решать в сферических координатах, поскольку поле обладает сферической симметрией

```
> SetCoordinates('spherical'[rho,theta,phi]);
```

spherical _{ρ, θ, ϕ}

```
> U:=1/rho;
```

$$U := \frac{1}{\rho}$$

```
> Gradient(U);
```

$$-\frac{1}{\rho^2} \bar{\mathbf{e}}_\rho$$

Задача 9.16. Доказать равенство $\text{grad}(\phi\psi) = \phi \text{grad}(\psi) + \psi \text{grad}(\phi)$, где $\phi(x, y, z)$ и $\psi(x, y, z)$ — скалярные функции.

Решение. Для доказательства равенства вычислим отдельно левую и правую части этого равенства и убедимся, что они совпадают. В системе Maple задача решается так:

```
> `left-hand member` :=
```

```
Gradient(phi(x,y,z)*psi(x,y,z), 'cartesian'[x,y,z]);
```

```
left-hand member :=  $\left( \left( \frac{\partial}{\partial x} \phi(x, y, z) \right) \psi(x, y, z) + \phi(x, y, z) \left( \frac{\partial}{\partial x} \psi(x, y, z) \right) \right) \bar{\mathbf{e}}_{\mathbf{x}}$  +
```

```
 $\left( \left( \frac{\partial}{\partial y} \phi(x, y, z) \right) \psi(x, y, z) + \phi(x, y, z) \left( \frac{\partial}{\partial y} \psi(x, y, z) \right) \right) \bar{\mathbf{e}}_{\mathbf{y}}$  +
```

```
 $\left( \left( \frac{\partial}{\partial z} \phi(x, y, z) \right) \psi(x, y, z) + \phi(x, y, z) \left( \frac{\partial}{\partial z} \psi(x, y, z) \right) \right) \bar{\mathbf{e}}_{\mathbf{z}}$ 
```

```
> `right-hand member` :=
```

```
psi(x,y,z)*Gradient(phi(x,y,z), 'cartesian'[x,y,z])+
```

```
phi(x,y,z)*Gradient(psi(x,y,z), 'cartesian'[x,y,z]);
```

$$\begin{aligned}
\text{right-hand member} &:= \left(\left(\frac{\partial}{\partial x} \phi(x, y, z) \right) \psi(x, y, z) + \phi(x, y, z) \left(\frac{\partial}{\partial x} \psi(x, y, z) \right) \right) \bar{\mathbf{e}}_x \\
&+ \left(\left(\frac{\partial}{\partial y} \phi(x, y, z) \right) \psi(x, y, z) + \phi(x, y, z) \left(\frac{\partial}{\partial y} \psi(x, y, z) \right) \right) \bar{\mathbf{e}}_y + \\
&\left(\left(\frac{\partial}{\partial z} \phi(x, y, z) \right) \psi(x, y, z) + \phi(x, y, z) \left(\frac{\partial}{\partial z} \psi(x, y, z) \right) \right) \bar{\mathbf{e}}_z \\
> \text{left-hand member} - \text{right-hand member} ; \\
&0 \bar{\mathbf{e}}_x
\end{aligned}$$

Векторное поле

Если каждой точке M области Ω поставлен в соответствие вполне определенный вектор $\bar{\mathbf{a}}(M)$, то говорят, что в области Ω определено *векторное поле*.

Векторное поле $\bar{\mathbf{a}}(M)$ задается в декартовой системе координат тремя функциями $P(x, y, z)$, $Q(x, y, z)$ и $R(x, y, z)$, определенными в области Ω

$$\bar{\mathbf{a}}(M) = P(x, y, z)\bar{\mathbf{i}} + Q(x, y, z)\bar{\mathbf{j}} + R(x, y, z)\bar{\mathbf{k}}.$$

Предполагается, что эти функции имеют непрерывные частные производные во всей области Ω .

Для плоского векторного поля

$$\bar{\mathbf{a}}(M) = P(x, y)\bar{\mathbf{i}} + Q(x, y)\bar{\mathbf{j}}.$$

Векторной линией векторного поля $\bar{\mathbf{a}}(M)$ называется такая линия L , в каждой точке которой вектор $\bar{\mathbf{a}}(M)$ является касательным к этой линии. Через каждую точку векторного поля проходит одна векторная линия, если $|\bar{\mathbf{a}}(M)| \neq 0$. Множество всех векторных линий определяется как решение системы обыкновенных дифференциальных уравнений

$$\frac{dx}{P(x, y, z)} = \frac{dy}{Q(x, y, z)} = \frac{dz}{R(x, y, z)}.$$

В случае плоского векторного поля семейство векторных линий определяется системой обыкновенных дифференциальных уравнений

$$\frac{dx}{P(x, y)} = \frac{dy}{Q(x, y)}, \quad dz = 0.$$

Приведем несколько примеров векторных полей.

В гидродинамике с векторным полем встречаются, изучая поток жидкости, протекающей через какую-либо часть пространства. Поле скоростей движущихся частиц

жидкости является векторным полем $\vec{v}(x, y, z, t)$, в общем случае — нестационарным. Оно полностью характеризует рассматриваемый поток. В поле стационарного потока жидкости векторными линиями (*линиями тока*) являются траектории, описываемые частицами жидкости.

Поле сил тяготения $\vec{F}(x, y, z)$ или гравитационное поле является примером стационарного векторного поля. Векторные линии в силовом физическом поле называются также *силовыми линиями*.

Изучая скалярное поле потенциала $u(x, y, z)$, мы рассматривали в каждой точке его вектор $\text{grad } u$, характеризующий скорость изменения потенциала поля. Поле $\text{grad } u$ является векторным полем, построенным над скалярным полем $u(x, y, z)$.

Рассматривая поле $\text{grad } u$, где $u = \frac{q}{r}$ есть потенциал электростатического поля, мы получаем поле напряженности электростатического поля

$$\vec{E} = -\text{grad} \left(\frac{q}{r} \right) = \frac{q\vec{r}_0}{r^2}, \quad \vec{r}_0 = \frac{\vec{r}}{r}.$$

В поле напряженности электростатического поля $\vec{E} = \frac{q\vec{r}_0}{r^2}$ векторные линии — это лучи, исходящие из точки, в которой помещен заряд, и направленные в сторону убывания потенциала, так как $\vec{E} = -\text{grad } u$.

Если течение жидкости потенциально, то есть вектор скорости \vec{v} можно представить в виде $\vec{v} = \text{grad } \varphi$, где $\varphi(x, y, z)$ — скалярный потенциал скорости, то уравнения семейства линий тока можно записать в виде

$$\frac{dx}{\frac{\partial \varphi}{\partial x}} = \frac{dy}{\frac{\partial \varphi}{\partial y}} = \frac{dz}{\frac{\partial \varphi}{\partial z}}.$$

Примеры

Задача 9.17. Плоский поток жидкости характеризуется вектором скорости $\vec{v} = xy\vec{i} - x(2x+1)\vec{j}$. Построить векторные линии поля.

Решение. Дифференциальное уравнение векторных линий поля скоростей имеет вид

$$\frac{dx}{xy} = -\frac{dy}{x(2x+1)}.$$

Это уравнение с разделяющимися переменными, которое легко интегрируется

$$\int (2x+1) dx = -\int y dy + C \Rightarrow x^2 + x + \frac{y^2}{2} = C.$$

Или, выделяя полный квадрат в левой части последнего равенства

$$\left(x + \frac{1}{2}\right)^2 + \frac{y^2}{2} = C + \frac{1}{4}.$$

Таким образом, векторные линии поля скоростей $\vec{v} = xy\vec{i} - x(2x+1)\vec{j}$ есть эллипсы с центром в точке $\left(-\frac{1}{2}, 0\right)$ и полуосями $\sqrt{C + \frac{1}{4}}$ и $\sqrt{2\left(C + \frac{1}{4}\right)}$:

$$\frac{\left(x + \frac{1}{2}\right)^2}{C + \frac{1}{4}} + \frac{y^2}{2\left(C + \frac{1}{4}\right)} = 1.$$

Рассмотрим решение этой задачи в Maple. Как мы только что видели, нам придется решить обыкновенное дифференциальное уравнение. С командой `dsolve`, решающей обыкновенные дифференциальные уравнения и системы, мы уже знакомы по главе 2.

Итак, определим наше дифференциальное уравнение векторных линий поля скоростей в Maple. Напомним, что при составлении уравнений для указания производной применяются команды `diff` и `D`, а для обозначения производной в начальных и краевых условиях используется команда `D`. В нашем случае имеем

```
> ode := diff(y(x), x) = -(2*x+1)/y(x);
```

$$ode := \frac{d}{dx} y(x) = -\frac{2x+1}{y(x)}$$

Теперь для решения этого уравнения используем команду `dsolve`

```
> dsolve(ode, y(x), implicit);
```

$$y(x)^2 + 2x^2 + 2x - _C1 = 0$$

Третий аргумент `implicit` команды `dsolve` указывает системе, что решение уравнения следует искать в неявной форме. Поскольку мы не задали никаких дополнительных условий, наше решение содержит произвольную константу интегрирования (она обозначена системой как `_C1`).

Посмотрим, как выглядят векторные линии нашего поля скоростей. Для этого зададимся какими-нибудь значениями произвольной постоянной и нарисуем векторные линии с помощью следующих команд

```
> \_C1 := 0.1:
```

```
p1 := plots[implicitplot](eq, x=-2..1, y=-2..2, color=black):
```


```
\_C1 := 0.5:
```

```
p2 := plots[implicitplot](eq, x=-2..1, y=-2..2, color=brown,
numpoints=2000):
```

```

_C1:=0.9:
p3:=plots[implicitplot](eq,x=-2..1,y=-2..2,color=blue,
numpoints=2000):
_C1:=1.4:
p4:=plots[implicitplot](eq,x=-2..1,y=-2..2,
numpoints=2000):
plots[display]({p1,p2,p3,p4});

```




Команда `implicitplot` рисует двумерный график функции, заданной неявно уравнением. Эта команда находится в специальном пакете `plots`. Таким образом, с помощью команды `implicitplot` мы подготовили четыре графика `p1`, `p2`, `p3`, `p4` для четырех различных значений произвольной постоянной интегрирования, а затем с помощью команды `display`, которая также находится в пакете `plots`, мы отобразили эти графики на одном рисунке.

Для визуализации поля направлений можно воспользоваться командой `dfieldplot` находящейся в пакете `DEtools` и предназначенной для изображения векторного поля (поля направлений). Например, в нашем случае

```

> DEtools[dfieldplot](ode,y(x),x=-2..1,y=-1..1,
arrows=MEDIUM,color=blue);

```



Обратите внимание, здесь, чтобы отобразить векторное поле направлений, нам не пришлось предварительно решать дифференциальное уравнение.

Задача 9.18. Поток несжимаемой жидкости имеет потенциал $\phi(x, y) = \frac{x}{x^2 + y^2}$.

Найти траектории движения частиц жидкости.

Решение. Подключим пакет

```
> with(VectorCalculus):
```

```
Warning, the assigned names <,> and <|> now have a global binding
```

```
Warning, these protected names have been redefined and unprotected: *, +, ., Vector, diff, int, limit, series
```

Определим потенциал

```
> phi:=x/(x^2+y^2);
```

$$\phi := \frac{x}{x^2 + y^2}$$

Находим векторное поле скоростей как градиент потенциала

```
> grad:=simplify(Gradient(phi, 'cartesian'[x,y,z]));
```

$$\mathit{grad} := -\frac{x^2 - y^2}{(x^2 + y^2)^2} \mathbf{e}_x - \frac{2xy}{(x^2 + y^2)^2} \mathbf{e}_y$$

Определяем теперь дифференциальное уравнение векторных линий поля скоростей

```
> ode:=diff(y(x),x)=simplify(subs(y=y(x),grad[2]/grad[1]));
```

$$\mathit{ode} := \frac{d}{dx} y(x) = \frac{2xy(x)}{x^2 - y(x)^2}$$

и решаем это уравнение

```
> simplify(dsolve(ode,y(x),implicit),symbolic);
```

$$\ln(y(x)) - \ln(x^2 + y(x)^2) - _C1 = 0$$

Чтобы получить уравнения траекторий удобно переобозначить константу интегрирования

```
> subs(_C1=ln(C), %);
```

$$\ln(y(x)) - \ln(x^2 + y(x)^2) - \ln(C) = 0$$

Теперь найдем уравнения траекторий

```
> eq:=C=subs(y(x)=y, solve(%, C));
```

$$eq := C = \frac{y}{x^2 + y^2}$$

Это, очевидно, окружности с центром на оси Oy , проходящие через начало координат. Чтобы убедиться в этом, зададим какие-нибудь значения постоянной интегрирования и отобразим траектории на одном графике

```
> C:=1:
```

```
p1:=plots[implicitplot](eq, x=-1..1, y=-0.1..1,  
color=black):
```

```
C:=2:
```

```
p2:=plots[implicitplot](eq, x=-1..1, y=-0.1..1,  
color=brown, numpoints=2000):
```


```
C:=3:
```

```
p3:=plots[implicitplot](eq, x=-1..1, y=-0.1..1,  
color=blue, numpoints=2000):
```

```
C:=4:
```


```
p4:=plots[implicitplot](eq, x=-1..1, y=-0.1..1,  
numpoints=2000):
```

```
plots[display]({p1, p2, p3, p4});
```



Поле направлений можно посмотреть с помощью команды

```
> DEtools[dfieldplot](ode, y(x), x=-10..10, y=-10..10,
arrows=MEDIUM, color=blue);
```



Поток векторного поля. Формула Остроградского

Потоком Π векторного поля $\vec{a}(M)$ через поверхность S называется поверхностный интеграл

$$\Pi = \iint_S \vec{a} \cdot \vec{n} dS,$$

где $\vec{n} = \vec{i} \cos \alpha + \vec{j} \cos \beta + \vec{k} \cos \gamma$ — единичный вектор нормали к поверхности в ее текущей точке M . Поверхность должна быть ориентированной; это означает, что в каждой точке поверхности выбрано одно из двух направлений нормали, так чтобы вектор $\vec{n}(M)$ менялся на поверхности непрерывно. В случае замкнутой поверхности в качестве вектора $\vec{n}(M)$ мы всегда будем выбирать вектор *внешней* нормали.

Если векторное поле имеет вид

$$\vec{a} = P(x, y, z)\vec{i} + Q(x, y, z)\vec{j} + R(x, y, z)\vec{k},$$

то поток выражается интегралом

$$\Pi = \iint_S [P(x, y, z) \cos \alpha + Q(x, y, z) \cos \beta + R(x, y, z) \cos \gamma] dS.$$

Для вычисления этого интеграла удобно проектировать поверхность S на одну из координатных плоскостей. Пусть, например, поверхность S взаимно однозначно проектируется на плоскость xOy . Тогда будем иметь

$$dS = \frac{dxdy}{|\cos \gamma|}$$

и

$$\Pi = \iint_{S_{xy}} \frac{\vec{a} \cdot \vec{n}}{|\cos \gamma|} dx dy, \quad (9.7)$$

где S_{xy} — проекция поверхности S на плоскость xOy . Чтобы вычислить этот последний интеграл, записывают уравнение поверхности S в виде $z = f(x, y)$ и подставляют это выражение вместо z в подынтегральную функцию.

В общем случае произвольной поверхности S разбивают ее на куски S_1, S_2, \dots, S_m так, чтобы каждый кусок взаимно однозначно проектировался на одну из координатных плоскостей. Тогда

$$\Pi = \iint_S \vec{a} \cdot \vec{n} dS = \sum_{i=1}^m \iint_{S_i} \vec{a} \cdot \vec{n} dS,$$

где каждый из интегралов справа в последней формуле вычисляется, как указано выше.

Пусть S — замкнутая поверхность, ограничивающая объем V . Если векторное поле $\vec{a}(M)$ определено и непрерывно дифференцируемо во всем объеме V , включая его границу S , то справедлива формула Остроградского

$$\iint_S [P \cos \alpha + Q \cos \beta + R \cos \gamma] dS = \iiint_V \left[\frac{\partial P}{\partial x} + \frac{\partial Q}{\partial y} + \frac{\partial R}{\partial z} \right] dx dy dz.$$

В левой части стоит поток Π поля \vec{a} через поверхность S ; таким образом,

$$\Pi = \iiint_V \left[\frac{\partial P}{\partial x} + \frac{\partial Q}{\partial y} + \frac{\partial R}{\partial z} \right] dx dy dz. \quad (9.8)$$

Формула (9.8) во многих случаях упрощает вычисление потока через замкнутую поверхность S .

Если поверхность S_1 не замкнута, то иногда следует ее дополнить до замкнутой поверхности S с помощью некоторой поверхности S_2 . По свойству аддитивности интегралов, очевидно, будем иметь

$$\Pi_S = \Pi_{S_1} + \Pi_{S_2} \Rightarrow \Pi_{S_1} = \Pi_S - \Pi_{S_2}.$$

Это дает возможность вычислять поток через поверхность S_1 , если известны потоки Π_S и Π_{S_2} . Для вычисления потока Π_S через замкнутую поверхность S можно воспользоваться формулой Остроградского.

Примеры

Задача 9.19. Вычислить поток вектора $\vec{a} = (x - 2z)\vec{i} + (3z - 4x)\vec{j} + (5x + y)\vec{k}$ через треугольник ABC с вершинами $A(1, 0, 0)$, $B(0, 1, 0)$ и $C(0, 0, 1)$ (рис. 9.1).




Рис. 9.1. К задаче 9.19

Решение. Будем решать задачу в системе Maple. Наметим план решения. Искомый поток вычисляем по формуле (9.7). Уравнение плоскости треугольника ABC имеет вид

$$F(x, y, z) = x + y + z - 1 = 0.$$

Отсюда, по формуле

$$\vec{n} = \frac{\text{grad } F}{|\text{grad } F|},$$

найдем выражение для вектора нормали к этой плоскости. Затем вычислим подынтегральное выражение в формуле (9.7). После этого поток Π вычисляем как двойной интеграл по площади треугольника $A'B'C'$, являющегося проекцией треугольника ABC на плоскость xOy . Очевидно, $A' = (1, 0)$, $B' = (0, 1)$ и $C' = (0, 0)$.

Реализуем теперь наш план в системе Maple. Подключаем пакет VectorCalculus.

```
> with(VectorCalculus):
```

```
Warning, the assigned names <,> and <|> now have a global binding
```

```
Warning, these protected names have been redefined and unprotected: *, +, ., D, Vector, diff, int, limit, series
```

Определяем заданное векторное поле

```
> a := VectorField(<x-2*z, 3*z-4*x, 5*x+y>, 'cartesian'[x,y,z]);
```

$$a := (x - 2z) \vec{e}_x + (3z - 4x) \vec{e}_y + (5x + y) \vec{e}_z$$

Задаем плоскость треугольника

```
> x+y+z-1=0; F:=x+y+z-1;
```

$$x + y + z - 1 = 0$$

$$F := x + y + z - 1$$

и определяем вектор нормали к этой плоскости

```
> g:=Gradient(F, 'cartesian'[x,y,z]);
mod_g:=sqrt(g[1]^2+g[2]^2+g[3]^2);
n:=g/mod_g;
```

$$g := \bar{e}_x + \bar{e}_y + \bar{e}_z$$

$$mod_g := \sqrt{3}$$

$$n := \frac{\sqrt{3}}{3} \bar{e}_x + \frac{\sqrt{3}}{3} \bar{e}_y + \frac{\sqrt{3}}{3} \bar{e}_z$$

Таким образом, направляющие косинусы нормали будут

```
> 'cos(alpha)'=n[1]; 'cos(beta)'=n[2]; 'cos(gamma)'=n[3];
```

$$\cos(\alpha) = \frac{\sqrt{3}}{3}$$

$$\cos(\beta) = \frac{\sqrt{3}}{3}$$

$$\cos(\gamma) = \frac{\sqrt{3}}{3}$$

Вычисляем теперь подынтегральную функцию

```
> isolate(F, z); assign(%); f:=simplify(a.n/n[3]);
z = -x - y + 1
f := x + 1
```

Итак, искомый поток выражается через следующий двойной интеграл

$$\Pi = \int_0^1 \int_0^{1-x} (x+1) dx dy .$$

Этот последний интеграл можно вычислить в системе Maple несколькими способами. Один из них — непосредственное вычисление двойного интеграла через повторный.

```
> PI:=
Int(Int(f, y=0..1-x), x=0..1)=int(int(f, y=0..1-x), x=0..1);
```

$$\Pi := \int_0^1 \int_0^{1-x} x + 1 dy dx = \frac{2}{3}$$

Искомый поток можно вычислить в пакете VectorCalculus с помощью команд `int` или `Flux`. Рассмотрим эти возможности.

Вычисление с помощью команды `int`. Формат команды остается прежним: `int(f, dom, inert)`. Теперь эта команда вычисляет интеграл от функции `f` по области `dom`. Последний, третий, аргумент команды `inert` — указывает на необходимость возвращения интеграла в инертной (отложенной, не вычисленной) форме (`Int`). Если задан третий аргумент `inert`, то интеграл не вычисляется, а выводится его изображение (формула). Эта возможность реализована только в версии Maple, начиная с девятой и, конечно, является очень полезной, так как позволяет проконтролировать формулу, по которой ведется расчет. Второй аргумент `dom` — задает область интегрирования. Возможные области интегрирования:

- `Circle(cen, rad)` — окружность с заданным центром и радиусом;
- `Ellipse(eq)` — эллипс, заданный уравнением;
- `Parallelepiped(r1, [r2, r3, ...])` — параллелепипед (каждое `ri` задается как `algebraic .. algebraic`)
- `Regtangle(r1, r2)` — определяет параллелепипед в двумерном евклидовом пространстве;
- `Region(r1, [r2, r3, ...])` — каждое `ri` задается как `algebraic .. algebraic` и i -ая переменная может зависеть от $(i-1)$ переменных; интеграл по области может быть легко преобразован к повторному интегралу.
- `Sector(obj, start, finish)` — параметр `obj` может быть `Circle` или `Ellipse` структурой; параметры `start, finish` задаются как `algebraic .. algebraic`;
- `Sphera(cen, rad)` — сфера с заданным центром и радиусом;
- `Tetrahedron(v1, v2, v3, v4)` — тетраэдр (четырёхгранник) с заданными вершинами;
- `Triangle(v1, v2, v3)` — треугольник с заданными вершинами.

Вернемся к нашей задаче. Нам надо вычислить интеграл от функции $f(x) = x + 1$ по треугольнику с заданными вершинами

```
> int(f, [x, y]=Triangle(<0, 0>, <0, 1>, <1, 0>), 'inert')=
int(f, [x, y]=Triangle(<0, 0>, <0, 1>, <1, 0>));
```

$$\int_0^1 \int_0^{-x+1} x + 1 \, dy \, dx = \frac{2}{3}$$

Здесь мы специально, для контроля правильности расчета, вывели начертание вычисляемого интеграла (формулу). Как видим, это как раз тот интеграл, который нам надо вычислить.

Еще одна возможность вычисления потока вектора — использование команды `Flux`. В этом случае будем иметь

```
> z := 'z':
```

```
Flux(VectorField(<x-2*z,3*z-4*x,5*x+y>,cartesian[x,y,z]),
Surface(<s,t,1-s-t>,[s,t]=Rectangle(0..1-t,0..1)),
'inert')=
Flux(VectorField(<x-2*z,3*z-4*x,5*x+y>,cartesian[x,y,z]),
Surface(<s,t,1-s-t>,[s,t]=Rectangle(0..1-t,0..1)));
```

$$\int_0^1 \int_0^{1-t} s + 1 \, ds \, dt = \frac{2}{3}$$

Здесь мы также воспользовались возможностью инертной формы, чтобы проконтролировать расчетную формулу. Преимущества этого способа в том, что задача решается сразу, одной командой, без предварительных вычислений, как это было в предыдущих случаях (не требуется вычислять нормаль к поверхности, подинтегральную функцию и т. д.).

Итак, окончательно, искомый поток $\Pi = \frac{2}{3}$.

Кратко прокомментируем применение команды Flux. Формат команды

```
Flux(f,dom,inert);
```

Команда Flux(f,dom) вычисляет поток векторного поля f через поверхность в трехмерном евклидовом пространстве, заданную как dom. Поверхность может быть определена следующими структурами: Box(r1,r2,r3,[dir]) — параллелепипед (каждое ri задается как algebraic .. algebraic); Sphere(cen,rad,[dir]) — сфера с заданным центром и радиусом; Surface(v) — поверхность. Аргумент (не обязательный) dir указывает направление нормали (внутреннее inward или внешнее outward). Более подробную информацию и примеры, как всегда, можно получить по справке, набрав в строке ?Flux и нажав клавишу ввода.

Задача 9.20. Вычислить поток вектора $\vec{a} = x^2\vec{i} + x\vec{j} + xz\vec{k}$ через замкнутую поверхность S, состоящую из поверхности параболоида вращения $S_1: y = x^2 + z^2$ и из плоскостей S_2, S_3, S_4 , заданных соответственно уравнениями $y = 1, x = 0, z = 0$ (рис. 9.2).




Рис. 9.2. К задаче 9.20

Решение. Искомый поток, очевидно, можно представить как сумму потоков через поверхности S_1, S_2, S_3, S_4 , то есть $\Pi = \Pi_1 + \Pi_2 + \Pi_3 + \Pi_4$. Решаем задачу в Maple.

```
> with(VectorCalculus):
```

```
Warning, the assigned names <, > and <|> now have a global binding
```

```
Warning, these protected names have been redefined and unprotected: *, +, ., D, Vector, diff, int, limit, series
```

Задаем векторное поле

```
> a:=VectorField(<x^2,x,x*z>,'cartesian'[x,y,z]);
```

$$a := x^2 \bar{\mathbf{e}}_x + x \bar{\mathbf{e}}_y + xz \bar{\mathbf{e}}_z$$

Найдем поток через поверхность S_1 . Уравнение поверхности

```
> x^2+z^2-y=0; F:=x^2+z^2-y;
```

$$x^2 + z^2 - y = 0$$

$$F := x^2 + z^2 - y$$

Отсюда находим выражение для вектора нормали

```
> g:=Gradient(F,'cartesian'[x,y,z]);
```

```
mod_g:=sqrt(g[1]^2+g[2]^2+g[3]^2);
```

```
n:=g/mod_g;
```

$$g := 2x \bar{\mathbf{e}}_x - \bar{\mathbf{e}}_y + 2z \bar{\mathbf{e}}_z$$

$$\text{mod}_g := \sqrt{4x^2 + 1 + 4z^2}$$

$$n := \frac{2x}{\sqrt{4x^2 + 1 + 4z^2}} \bar{\mathbf{e}}_x - \frac{1}{\sqrt{4x^2 + 1 + 4z^2}} \bar{\mathbf{e}}_y + \frac{2z}{\sqrt{4x^2 + 1 + 4z^2}} \bar{\mathbf{e}}_z$$

Итак, вектор единичной нормали $\vec{n} = \frac{2x\vec{i} - \vec{j} + 2z\vec{k}}{\sqrt{4x^2 + 4z^2 + 1}}$. Таким образом, направляющие косинусы нормали будут

> 'cos(alpha)'=n[1]; 'cos(beta)'=n[2]; 'cos(gamma)'=n[3];

$$\cos(\alpha) = \frac{2x}{\sqrt{4x^2 + 1 + 4z^2}}$$

$$\cos(\beta) = -\frac{1}{\sqrt{4x^2 + 1 + 4z^2}}$$

$$\cos(\gamma) = \frac{2z}{\sqrt{4x^2 + 1 + 4z^2}}$$

Вычисляем подынтегральную функцию

> isolate(F,x); assign(%); f:=simplify(a.n/n[1]);

$$x = \text{RootOf}(-Z^2 + z^2 - y)$$

$$f := -\frac{1}{2} + y$$

Итак, подынтегральная функция $\frac{\vec{a} \cdot \vec{n}}{|\cos \alpha|} = y - \frac{1}{2}$. Следовательно, искомый поток Π_1

будет

> int(f,[y,z]=Region(0..1,0..sqrt(y)), 'inert')=

int(f,[y,z]=Region(0..1,0..sqrt(y)));

PI[1]:=rhs(%);

$$\int_0^1 \int_0^{\sqrt{y}} -\frac{1}{2} + y \, dz \, dy = \frac{1}{15}$$

$$\Pi_1 := \frac{1}{15}$$

Таким образом, мы получили

$$\Pi_1 = \iint_{S_1} \frac{\vec{a} \cdot \vec{n}}{|\cos \alpha|} \, dy \, dz = \int_0^1 \int_0^{\sqrt{y}} \left(y - \frac{1}{2} \right) \, dy \, dz = \frac{1}{15}.$$

Найдем поток Π_2 через поверхность S_2 . Внешней нормалью к поверхности S_2 является единичный орт оси Oy , то есть

> x:='x': 'abs(cos(beta))'=1;

j:=VectorField(<0,1,0>,'cartesian'[x,y,z]);

$$|\cos(\beta)| = 1$$

$$j := \frac{\vec{e}}{y}$$

Подынтегральная функция в этом случае будет

$$> g := a \cdot j;$$

$$g := x$$

Итак, мы получили $\cos\beta = 1$, $\frac{\vec{a} \cdot \vec{n}}{|\cos\beta|} = x$. Таким образом, поток Π_2 будет

```
> int(g, [x, z]=Sector(Circle(<0,0>,1), 0, Pi/2), 'inert') =
int(x, [x, z]=Sector(Circle(<0,0>,1), 0, Pi/2));
PI[2] := rhs(%);
```

$$\int_0^1 \int_0^{\frac{\pi}{2}} x^2 \cos(z) dz dx = \frac{1}{3}$$

$$\Pi_2 := \frac{1}{3}$$

Найдем теперь потоки Π_3 и Π_4 . На плоскости $x = 0$ заданный вектор поля равен нулю. Следовательно, $\Pi_3 = 0$. Далее, на плоскости $z = 0$ имеем $\vec{a} = x^2 \vec{i} + x \vec{j}$. Внешней нормалью к поверхности S_4 является вектор $\vec{n} = -\vec{k}$. Следовательно, $\vec{a} \cdot \vec{n} = 0$, и поэтому $\Pi_4 = 0$.

Окончательно, для полного потока находим

$$> PI := PI[1] + PI[2];$$

$$\Pi := \frac{2}{5}$$

Итак, полный поток через замкнутую поверхность $\Pi = \frac{2}{5}$.

Задача 9.21. Найти поток векторного поля $\vec{a} = (x - z)\vec{i} + (z^2 - y^2)\vec{j} + (x + z)\vec{k}$ через часть цилиндрической поверхности $S: R^2 = x^2 + z^2$, ограниченную плоскостями $y = 0$ и $y = b$, $b > 0$ (рис. 9.3).




Рис. 9.3. К задаче 9.21

Решение.

```
> with(VectorCalculus):
```

```
Warning, the assigned names <,> and <|> now have a global binding
```

```
Warning, these protected names have been redefined and unprotected: *, +, .., D, Vector, diff, int, limit, series
```

Задаем векторное поле

```
> a:=VectorField(<x-z, z^2-y^2, x+z>, 'cartesian'[x,y,z]);
```

$$a := (x - z) \bar{\mathbf{e}}_x + (z^2 - y^2) \bar{\mathbf{e}}_y + (x + z) \bar{\mathbf{e}}_z$$

Задача решается с помощью следующей команды:

```
> assume(R>0):
```

```
> 2*Flux(a, Surface(<sqrt(R^2-z^2), y, z>,
[y,z]=Rectangle(0..b, -R..R)), 'inert') =
2*Flux(a, Surface(<sqrt(R^2-z^2), y, z>,
[y,z]=Rectangle(0..b, -R..R)));
```

```
'PI'=rhs(%);
```

$$2 \int_{-R}^R \int_0^b \left(\sqrt{R^2 - z^2} - z + \frac{(\sqrt{R^2 - z^2} + z)z}{\sqrt{R^2 - z^2}} \right) dy dz = 2 b R^2 \pi$$

$$\Pi = 2 b R^2 \pi$$

Здесь мы поставили множитель 2, так как полный поток равен сумме потоков через части цилиндрической поверхности, расположенные по одну и другую сторону коор-

динатной плоскости yOz (при $x > 0$ и при $x < 0$). В силу симметрии эти потоки равны.

Можно выполнить все вычисления по формуле

$$\Pi = \iint_{S_{yz}} \frac{\vec{a} \cdot \vec{n}}{|\cos \alpha|} dydz.$$

с помощью команды `int`.

Для этого вычислим подынтегральное выражение. Сначала найдем вектор единичной нормали к цилиндрической поверхности

```
> F:=x^2+z^2-R^2;
g:=Gradient(F, 'cartesian'[x,y,z]);
mod_g:=sqrt(g[1]^2+g[2]^2+g[3]^2);
n:=g/mod_g;
```

$$F := x^2 + z^2 - R^2$$

$$g := 2x \vec{e}_x + 2z \vec{e}_z$$

$$mod_g := 2\sqrt{x^2 + z^2}$$

$$n := \frac{x}{\sqrt{x^2 + z^2}} \vec{e}_x + \frac{z}{\sqrt{x^2 + z^2}} \vec{e}_z$$

Итак, вектор нормали $\vec{n} = \frac{x\vec{i} + z\vec{k}}{\sqrt{x^2 + z^2}}$.

Теперь вычисляем подынтегральную функцию

```
> sol:=solve(F,x);
sol:=sqrt(-z^2+R^2), -sqrt(-z^2+R^2)
> x:=sol[1];
x:=sqrt(-z^2+R^2)
> f:=simplify(a.n/n[1]);
f:=R^2/sqrt(-z^2+R^2)
```

Итак, $\frac{\vec{a} \cdot \vec{n}}{|\cos \alpha|} = \frac{R^2}{\sqrt{R^2 - z^2}}$.

Искомый поток выражается удвоенным интегралом по части цилиндрической поверхности, расположенной по одну сторону координатной плоскости yOz (при $x > 0$).

```
> 2*int( f, [y,z] = Rectangle( 0..b, -R..R ), 'inert' );
```

$$2 \int_{-R}^R \int_0^b \frac{R^2}{\sqrt{-z^2 + R^2}} dy dz$$

```
> simplify(value(%));
```

$$2 \pi R^2 b$$

Итак, искомый поток $\Pi = 2\pi bR^2$.

Задача 9.22. Найти поток векторного поля $\vec{a} = x\vec{i} + (y+z)\vec{j} + (z-y)\vec{k}$ через часть сферической поверхности $S: x^2 + y^2 + z^2 = 9$, ограниченную плоскостью $z = \frac{3}{2}$ (рис. 9.4).




Рис. 9.4. К задаче 9.22

Решение.

```
> with(VectorCalculus):
```

Warning, the assigned names <, > and <|> now have a global binding

Warning, these protected names have been redefined and unprotected: *, +, .., D, Vector, diff, int, limit, series

Задаем векторное поле

```
> a:=VectorField(<x,y+z,z-y>, 'cartesian'[x,y,z]);
```

$$a := x \vec{e}_x + (y+z) \vec{e}_y + (z-y) \vec{e}_z$$

Находим поток с помощью команды

```
> Flux(a, Surface(<3,s,t>, s=0..Pi/3, t=0..2*Pi,
```

```

coords=spherical), 'inert')=
Flux(a, Surface(<3, s, t>, s=0..Pi/3, t=0..2*Pi,
coords=spherical));

```

$$\int_0^{\frac{\pi}{3}} \int_0^{2\pi} 27 \sin(s)^3 \cos(t)^2 + 9 (3 \sin(s) \sin(t) + 3 \cos(s)) \sin(s)^2 \sin(t) + (3 \cos(s) - 3 \sin(s) \sin(t)) (9 \cos(s) \cos(t)^2 \sin(s) + 9 \cos(s) \sin(t)^2 \sin(s)) dt ds = 27 \pi$$

Итак, искомый поток $\Pi = 27\pi$.

Задача 9.23. Найти поток векторного поля $\vec{a} = xy^2\vec{i} + yz^2\vec{j} + zx^2\vec{k}$ через полную сферическую поверхность $S : x^2 + y^2 + z^2 = R^2$.

Решение.

```
> with(VectorCalculus):
```

```
Warning, the assigned names <,> and <|> now have a global binding
```

```
Warning, these protected names have been redefined and unprotected: *, +, ., D, Vector, diff, int, limit, series
```

Задаем векторное поле

```
> a:=VectorField(<x*y^2, y*z^2, z*x^2>, 'cartesian'[x,y,z]);
```

$$a := x y^2 \vec{e}_x + y z^2 \vec{e}_y + z x^2 \vec{e}_z$$

Находим поток

```
> Flux(a, Sphere(<0,0,0>,R), 'inert')=
```

```
Flux(a, Sphere(<0,0,0>,R));
```

$$\int_0^{2\pi} \int_0^{\pi} R^5 (-1 + \cos(\phi)^2) \sin(\phi) (\cos(\theta)^2 - \cos(\theta)^4 - \cos(\phi)^2 \cos(\theta)^2 + \cos(\theta)^4 \cos(\phi)^2 + \cos(\phi)^2) d\phi d\theta = \frac{4 R^5 \pi}{5}$$

Итак, $\Pi = \frac{4\pi R^5}{5}$.

Задача 9.24. Найти поток векторного поля $\vec{a} = y\vec{i} - x\vec{j} + z\vec{k}$ через замкнутую поверхность, состоящую из поверхности конуса $x^2 + y^2 = z^2$ и плоскости $z = 1$.

Решение. Решим задачу с помощью формулы Остроградского.

```
> with(VectorCalculus):
```

Warning, the assigned names <, > and <|> now have a global binding

Warning, these protected names have been redefined and unprotected: *, +, ., D, Vector, diff, int, limit, series

Задаем векторное поле

```
> a:=VectorField(<y,-x,z>,'cartesian'[x,y,z]);
```

$$a := y \bar{\mathbf{e}}_x - x \bar{\mathbf{e}}_y + z \bar{\mathbf{e}}_z$$

Находим подынтегральную функцию в формуле (9.7)

```
> Diff(a[1],x)+Diff(a[2],y)+Diff(a[3],z)=
```

```
diff(a[1],x)+diff(a[2],y)+diff(a[3],z);f:=value(rhs(%));
```

$$\left(\frac{\partial}{\partial x} y\right) + \left(\frac{\partial}{\partial y} (-x)\right) + \left(\frac{d}{dz} z\right) = 1$$

$$f := 1$$

Интеграл (9.8) вычисляем в цилиндрических координатах

```
> Int(Int(Int(f*r,r=0..z),phi=0..2*Pi),z=0..1)=
```

```
int(int(int(f*r,r=0..z),phi=0..2*Pi),z=0..1);
```

```
PI:=value(rhs(%));
```

$$\int_0^1 \int_0^{2\pi} \int_0^z r \, dr \, d\phi \, dz = \frac{\pi}{3}$$

$$\Pi := \frac{\pi}{3}$$

Итак, $\Pi = \frac{\pi}{3}$. Геометрически этот поток равен объему конуса.

Дивергенция векторного поля

Пусть \vec{a} — векторное поле. Зафиксируем в этом поле какую-нибудь точку M и окружим ее произвольной замкнутой поверхностью S . Рассмотрим поток векторного поля \vec{a} через эту поверхность S

$$\Pi = \iint_S \vec{a} \cdot \vec{n} dS.$$

Одной из важнейших характеристик в теории поля является предел отношения потока Π к объему V , ограниченному поверхностью S , когда S стягивается в точку M . Этот предел называют *дивергенцией* (*расходимостью*) векторного поля \vec{a} в точке M и обозначают $\operatorname{div} \vec{a}(M)$. Таким образом,

$$\operatorname{div} \vec{a}(M) = \lim_{S \rightarrow M} \frac{\iint_S \vec{a} \cdot \vec{n} dS}{V}. \quad (9.9)$$

Предполагается, что указанный предел не зависит от того, какие формы принимает поверхность S в процессе стягивания ее в точку M .

Для существования предела (9.9) достаточно, чтобы поле \vec{a} было непрерывно дифференцируемым, то есть, чтобы проекции вектора \vec{a} имели непрерывные частные производные.

Если дивергенция поля \vec{a} в точке M положительна, то точка M называется *источником поля*; если дивергенция отрицательна, то точка M называется *стоком поля*.

Если векторное поле имеет вид $\vec{a} = a_x \vec{i} + a_y \vec{j} + a_z \vec{k}$, то его дивергенция вычисляется по формуле

$$\operatorname{div} \vec{a} = \frac{\partial a_x}{\partial x} + \frac{\partial a_y}{\partial y} + \frac{\partial a_z}{\partial z}. \quad (9.10)$$

Используя символический вектор Гамильтона ∇ (набла оператор), дивергенцию поля \vec{a} можно представить в виде скалярного произведения

$$\operatorname{div} \vec{a} = \nabla \cdot \vec{a}.$$

С учетом формул (9.10) и (9.8) можно записать выражение для потока векторного поля \vec{a} через замкнутую поверхность S следующим образом

$$\Pi = \iiint_V \operatorname{div} \vec{a} dV.$$

Таким образом, мы имеем векторную запись формулы Остроградского

$$\iint_S (\vec{a} \cdot \vec{n}) dS = \iiint_V (\nabla \cdot \vec{a}) dV.$$

Дивергенция обладает следующими очевидными свойствами:

1. $\operatorname{div}(c_1 \vec{a}_1 + c_2 \vec{a}_2) = c_1 \operatorname{div} \vec{a}_1 + c_2 \operatorname{div} \vec{a}_2$;
2. $\operatorname{div}(u \vec{a}) = u \operatorname{div} \vec{a} + \vec{a} \cdot \operatorname{grad} u$.

Здесь u — скалярное поле; \vec{a} , \vec{a}_1 , \vec{a}_2 — векторные поля; $c_1, c_2 = \text{const}$.

Примеры

Задача 9.25. Вычислить дивергенцию радиус-вектора \vec{r} .

Решение. Имеем: $\vec{r} = x\vec{i} + y\vec{j} + z\vec{k}$. Применяя формулу (9.10), находим

$$\operatorname{div} \vec{r} = \frac{\partial x}{\partial x} + \frac{\partial y}{\partial y} + \frac{\partial z}{\partial z} = 3.$$

Итак, $\operatorname{div} \vec{r} = 3$.

Задача 9.26. Вычислить дивергенцию векторного поля $\vec{a} = r^2 \vec{c}$, где \vec{c} — постоянный вектор.

Решение. Все вычисления выполним в Maple. Подключаем пакет векторных вычислений и определяем заданные поля

```
> with(VectorCalculus):
Warning, the assigned names <,> and <|> now have a global
binding
Warning, these protected names have been redefined and
unprotected: *, +, ., D, Vector, diff, int, limit, series
> r:=VectorField(<x,y,z>,'cartesian'[x,y,z]);
      r := x  $\vec{e}_x$  + y  $\vec{e}_y$  + z  $\vec{e}_z$ 
> c:=VectorField(<C[1],C[2],C[3]>,'cartesian'[x,y,z]);
      c := C1  $\vec{e}_x$  + C2  $\vec{e}_y$  + C3  $\vec{e}_z$ 
```

Находим модуль радиус-вектора \vec{r} по формуле $|\vec{r}| = \sqrt{\vec{r} \cdot \vec{r}}$

```
> mod_r:=sqrt(r.r);
      mod_r :=  $\sqrt{x^2 + y^2 + z^2}$ 
```

Находим векторное поле \vec{a}

```
> a:=mod_r^2*c;
      a := (x2 + y2 + z2) C1  $\vec{e}_x$  + (x2 + y2 + z2) C2  $\vec{e}_y$  + (x2 + y2 + z2) C3  $\vec{e}_z$ 
```

И, наконец, вычисляем дивергенцию этого поля

```
> Divergence(a);
      2 x C1 + 2 y C2 + 2 z C3
```

Отметим, что дивергенцию можно вычислить в Maple разными способами. Кроме использованной уже нами команды `Divergence` тот же результат мы получим и так

```
> Del.a;
      2 x C1 + 2 y C2 + 2 z C3
```

или так

```
> Nabla.a;
```

$$2x C_1 + 2y C_2 + 2z C_3$$

Присмотревшись внимательнее к полученному результату, видим, что это есть не что иное, как удвоенное скалярное произведение векторов \vec{r} и \vec{c} . Проверим это

$$> 2 * (r . c) ;$$

$$2x C_1 + 2y C_2 + 2z C_3$$

Итак, $\operatorname{div}(r^2 \vec{c}) = 2(\vec{r} \cdot \vec{c})$.

Задача 9.27. Вычислить дивергенцию центрально-симметричного векторного поля $\vec{a} = f(r)\vec{r}$, где $f(r)$ — произвольная скалярная функция.

Решение. Для вычисления дивергенции этого поля применим формулу

$$\operatorname{div}(u\vec{a}) = u \operatorname{div} \vec{a} + \vec{a} \cdot \operatorname{grad} u,$$

в которой положим $u = f(r)$, $\vec{a} = \vec{r}$.

Учитывая, что $\operatorname{div} \vec{r} = 3$ (Задача 9.25) и $\operatorname{grad} f(r) = \frac{f'(r)}{r} \vec{r}$ (см. формулу (9.6)), получим

$$\operatorname{div}[f(r)\vec{r}] = 3f(r) + rf'(r).$$

При выводе последней формулы учтено, что $\vec{r} \cdot \vec{r} = r^2$.

Рассмотрим решение этой задачи в Maple. Считаем, что пакет векторных вычислений уже подключен и определяем заданные поля

```
> r := VectorField(<x, y, z>, 'cartesian'[x, y, z]);
```

$$r := x \vec{e}_x + y \vec{e}_y + z \vec{e}_z$$

```
> mod_r := (r . r)^(1/2);
```

$$\operatorname{mod}_r := \sqrt{x^2 + y^2 + z^2}$$

```
> a := f(mod_r) * r;
```

$$a := f(\sqrt{x^2 + y^2 + z^2}) x \vec{e}_x + f(\sqrt{x^2 + y^2 + z^2}) y \vec{e}_y + f(\sqrt{x^2 + y^2 + z^2}) z \vec{e}_z$$

Вычисляем дивергенцию

```
> Divergence(a);
```

$$\frac{D(f)(\sqrt{x^2 + y^2 + z^2}) x^2}{\sqrt{x^2 + y^2 + z^2}} + 3 f(\sqrt{x^2 + y^2 + z^2}) + \frac{D(f)(\sqrt{x^2 + y^2 + z^2}) y^2}{\sqrt{x^2 + y^2 + z^2}} + \frac{D(f)(\sqrt{x^2 + y^2 + z^2}) z^2}{\sqrt{x^2 + y^2 + z^2}}$$

Упростим теперь полученный результат

```
> collect(simplify(%), D(f)(mod_r));
```

$$\sqrt{x^2 + y^2 + z^2} D(f)(\sqrt{x^2 + y^2 + z^2}) + 3 f(\sqrt{x^2 + y^2 + z^2})$$

Подставим в найденное выражение обозначение модуля радиус-вектора \vec{r} :
 $\sqrt{x^2 + y^2 + z^2} = r$

$$\begin{aligned} &> \text{subs}((x^2+y^2+z^2)^(1/2)='r', \%); \\ & \quad r D(f)(r) + 3 f(r) \end{aligned}$$

Итак, мы получили $\text{div}[f(r)\vec{r}] = 3f(r) + rf'(r)$.

Ротор векторного поля

Ротором (или *вихрем*) векторного поля $\vec{a} = P(x, y, z)\vec{i} + Q(x, y, z)\vec{j} + R(x, y, z)\vec{k}$ называется векторное поле $\text{rot } \vec{a}$, которое определяется по формуле

$$\text{rot } \vec{a} = \left(\frac{\partial R}{\partial y} - \frac{\partial Q}{\partial z} \right) \vec{i} + \left(\frac{\partial P}{\partial z} - \frac{\partial R}{\partial x} \right) \vec{j} + \left(\frac{\partial Q}{\partial x} - \frac{\partial P}{\partial y} \right) \vec{k}.$$

В компактной форме ротор можно записать с помощью определителя третьего порядка:

$$\text{rot } \vec{a} = \begin{vmatrix} \vec{i} & \vec{j} & \vec{k} \\ \frac{\partial}{\partial x} & \frac{\partial}{\partial y} & \frac{\partial}{\partial z} \\ P & Q & R \end{vmatrix}. \quad (9.11)$$

Для вычисления ротора этот определитель разлагается по элементам первой строки, причем получающиеся определители второго порядка формально вычисляются по обычным правилам, а именно

$$\begin{vmatrix} \frac{\partial}{\partial y} & \frac{\partial}{\partial z} \\ Q & R \end{vmatrix} = \frac{\partial R}{\partial y} - \frac{\partial Q}{\partial z}, \quad \begin{vmatrix} \frac{\partial}{\partial x} & \frac{\partial}{\partial z} \\ P & R \end{vmatrix} = \frac{\partial R}{\partial x} - \frac{\partial P}{\partial z}, \quad \begin{vmatrix} \frac{\partial}{\partial x} & \frac{\partial}{\partial y} \\ P & Q \end{vmatrix} = \frac{\partial Q}{\partial x} - \frac{\partial P}{\partial y}.$$

Ротор векторного поля может быть представлен в символической форме как векторное произведение символического вектора Гамильтона (набла оператора) ∇ и вектора \vec{a}

$$\text{rot } \vec{a} = \nabla \times \vec{a}.$$

Ротор обладает следующими очевидными свойствами:

1. $\text{rot}(c_1\vec{a}_1 + c_2\vec{a}_2) = c_1 \text{rot } \vec{a}_1 + c_2 \text{rot } \vec{a}_2$;
2. $\text{rot}(u\vec{a}) = u \text{rot } \vec{a} + \text{grad } u \times \vec{a}$.

Здесь u — скалярное поле; \vec{a} , \vec{a}_1 , \vec{a}_2 — векторные поля; $c_1, c_2 = \text{const}$.

Примеры

Задача 9.28. Вычислить ротор радиус-вектора \vec{r} .

Решение. Имеем: $\vec{r} = x\vec{i} + y\vec{j} + z\vec{k}$. Применяя формулу (9.11), находим

$$\operatorname{rot} \vec{r} = \begin{vmatrix} \vec{i} & \vec{j} & \vec{k} \\ \frac{\partial}{\partial x} & \frac{\partial}{\partial y} & \frac{\partial}{\partial z} \\ x & y & z \end{vmatrix} = \left(\frac{\partial z}{\partial y} - \frac{\partial y}{\partial z} \right) \vec{i} + \left(\frac{\partial x}{\partial z} - \frac{\partial z}{\partial x} \right) \vec{j} + \left(\frac{\partial y}{\partial x} - \frac{\partial x}{\partial y} \right) \vec{k} = 0.$$

Итак, $\operatorname{rot} \vec{r} = 0$.

Задача 9.29. Вычислить ротор векторного поля $\vec{a} = y^2\vec{i} - x^2\vec{j} + z^2\vec{k}$.

Решение. Выполним все вычисления в Maple. Считаем, что пакет векторных вычислений уже подключен. Задаем векторное поле \vec{a}

```
> a:=VectorField(<y^2,-x^2,z^2>,'cartesian'[x,y,z]);
      a := y2  $\vec{e}_x$  - x2  $\vec{e}_y$  + z2  $\vec{e}_z$ 
```

Ротор векторного поля в системе Maple можно вычислить несколькими эквивалентными способами:

```
> Curl(a);
      (-2 x - 2 y)  $\vec{e}_z$ 

> Del &x a;
      (-2 x - 2 y)  $\vec{e}_z$ 

> Nabla &x a;
      (-2 x - 2 y)  $\vec{e}_z$ 

> CrossProduct(Del,a);
      (-2 x - 2 y)  $\vec{e}_z$ 
```

Первый из рассмотренных вариантов — использование специальной функции `Curl` для вычисления ротора векторного поля; остальные варианты — непосредственное вычисление ротора как векторного произведения символического вектора набла на вектор поля. Обратите внимание — для вычисления векторного произведения двух векторов в системе Maple можно использовать оператор `&x`, который следует поставить между векторами – сомножителями обязательно через пробел. Еще одна возможность — воспользоваться специальной командой `CrossProduct`, вычисляющей векторное произведение двух векторов (в этом случае векторы выступают в роли ар-

гументов данной команды; причем порядок аргументов важен, как это имеет место в векторном произведении).

Задача 9.30. Вычислить ротор центрально-симметричного векторного поля $\vec{a} = f(r)\vec{r}$, где $f(r)$ — произвольная скалярная функция.

Решение. Для вычисления ротора этого поля применим формулу

$$\operatorname{rot}(u\vec{a}) = u \operatorname{rot} \vec{a} + \operatorname{grad} u \times \vec{a},$$

в которой положим $u = f(r)$, $\vec{a} = \vec{r}$.

Так как $\operatorname{rot} \vec{r} = 0$ (см. задачу 9.28) и $\operatorname{grad} f(r) = \frac{f'(r)}{r} \vec{r}$ (см. формулу (9.6)), то

$$\operatorname{grad}(f(r)) \times \vec{r} = \frac{f'(r)}{r} \vec{r} \times \vec{r} = 0.$$

Следовательно, $\operatorname{rot}[f(r)\vec{r}] = 0$.

В системе Maple эта задача решается, например, так

```
> r:=VectorField(<x,y,z>,'cartesian'[x,y,z]);
```

$$r := x \vec{e}_x + y \vec{e}_y + z \vec{e}_z$$

```
> mod_r:=(r.r)^(1/2);
```

$$\text{mod_r} := \sqrt{x^2 + y^2 + z^2}$$

```
> a:=f(mod_r)*r;
```

$$a := f(\sqrt{x^2 + y^2 + z^2}) x \vec{e}_x + f(\sqrt{x^2 + y^2 + z^2}) y \vec{e}_y + f(\sqrt{x^2 + y^2 + z^2}) z \vec{e}_z$$

```
> Curl(a);
```

$$0 \vec{e}_x$$

Задача 9.31. Вычислить ротор поля скоростей точек вращающегося абсолютно твердого тела.

Решение. Как известно из теоретической механики, скорость выражается формулой

$$\vec{v} = \vec{\omega} \times \vec{r},$$

где $\vec{\omega}$ — вектор угловой скорости. Считая, что ось Oz направлена по вектору $\vec{\omega}$, будем иметь: $\vec{\omega} = \omega \vec{k}$ и, следовательно,

$$\vec{v} = \vec{\omega} \times \vec{r} = \omega \vec{k} \times \vec{r} = \begin{vmatrix} \vec{i} & \vec{j} & \vec{k} \\ 0 & 0 & \omega \\ x & y & z \end{vmatrix} = -\omega y \vec{i} + \omega x \vec{j}.$$

Отсюда

$$\operatorname{rot} \vec{v} = \begin{vmatrix} \vec{i} & \vec{j} & \vec{k} \\ \frac{\partial}{\partial x} & \frac{\partial}{\partial y} & \frac{\partial}{\partial z} \\ -\omega y & \omega x & 0 \end{vmatrix} = 2\omega \vec{k} = 2\vec{\omega}.$$

Рассмотрим решение этой задачи в Maple. Определим все необходимые векторные поля: вектор-радиус точки

```
> r:=VectorField(<x,y,z>,'cartesian'[x,y,z]);  
r := x  $\vec{e}_x$  + y  $\vec{e}_y$  + z  $\vec{e}_z$ 
```

вектор угловой скорости

```
> omega:=VectorField(<0,0,Omega>,'cartesian'[x,y,z]);  
omega := Omega  $\vec{e}_z$ 
```

скорость

```
> v:=omega &x r;  
v := -Omega y  $\vec{e}_x$  + Omega x  $\vec{e}_y$ 
```

Теперь вычисляем ротор поля скоростей

```
> Curl(v);  
2 Omega  $\vec{e}_z$ 
```

Таким образом, $\operatorname{rot} \vec{v}$ равен удвоенному вектору угловой скорости: $\operatorname{rot} \vec{v} = 2\vec{\omega}$.

Линейный интеграл и циркуляция векторного поля. Формула Стокса

Линейным интегралом векторного поля $\vec{a}(M)$ вдоль линии L называется криволинейный интеграл

$$W = \int_L \vec{a}(M) \cdot \vec{\tau}(M) dl,$$

где $\vec{\tau}(M)$ — единичный касательный вектор к линии L в точке M , dl — дифференциал дуги (рис. 9.5).




Рис. 9.5. К определению линейного интеграла

Если L — кусочно-гладкая линия, состоящая из нескольких гладких кусков L_1, L_2, \dots, L_n , то линейный интеграл векторного поля $\vec{a}(M)$ вдоль такой линии представляет собой сумму линейных интегралов по каждому ее куску, то есть

$$W = \int_L \vec{a}(M) \cdot \vec{\tau}(M) dl = \sum_{i=1}^n \int_{L_i} \vec{a}(M) \cdot \vec{\tau}(M) dl.$$

Для вычисления линейного интеграла удобно представить его в другом виде. Так как $\vec{\tau} dl = d\vec{r}$, где $d\vec{r}$ — дифференциал радиус вектора \vec{r} , описывающего данную линию L , то

$$\int_L \vec{a} \cdot \vec{\tau} dl = \int_L \vec{a} \cdot d\vec{r}.$$

В прямоугольной декартовой системе координат имеем

$$d\vec{r} = dx\vec{i} + dy\vec{j} + dz\vec{k},$$

$$\vec{a} = P(x, y, z)\vec{i} + Q(x, y, z)\vec{j} + R(x, y, z)\vec{k},$$

поэтому

$$W = \int_L \vec{a} \cdot d\vec{r} = \int_L P(x, y, z) dx + Q(x, y, z) dy + R(x, y, z) dz.$$

В случае, когда линия L задана параметрическими уравнениями $x = x(t)$, $y = y(t)$, $z = z(t)$, линейный интеграл вычисляется по формуле

$$W = \int_{t_1}^{t_2} \{P[x(t), y(t), z(t)]x'(t) + Q[x(t), y(t), z(t)]y'(t) + R[x(t), y(t), z(t)]z'(t)\} dt.$$

Если линия L — плоская, то

$$W = \int_L \vec{a} \cdot d\vec{r} = \int_L P(x, y) dx + Q(x, y) dy.$$

или

$$W = \int_{t_1}^{t_2} \{P[x(t), y(t)]x'(t) + Q[x(t), y(t)]y'(t)\} dt.$$

В частности, если кривая L задана явным уравнением $y = f(x)$, $x \in [a, b]$, то

$$W = \int_a^b \{P[x, y(x)] + Q[x, y(x)]y'(x)\} dx.$$

Циркуляцией векторного поля называется линейный интеграл векторного поля $\vec{a}(M)$ по замкнутому пути L

$$C = \oint_L \vec{a} \cdot d\vec{r}.$$

Формула Стокса, применение которой часто упрощает вычисление циркуляции, записывается в виде

$$\oint_L \vec{a} \cdot d\vec{r} = \iint_{\sigma} \text{rot } \vec{a} \cdot \vec{n} dS,$$

где σ — произвольная кусочно-гладкая поверхность, ограниченная контуром L .

С учетом формулы Стокса циркуляция векторного поля $\vec{a}(M)$ по замкнутому контуру L , ограничивающему поверхность σ , в декартовой системе координат выражается интегралом

$$C = \iint_{\sigma} \left[\left(\frac{\partial R}{\partial y} - \frac{\partial Q}{\partial z} \right) \cos \alpha + \left(\frac{\partial P}{\partial z} - \frac{\partial R}{\partial x} \right) \cos \beta + \left(\frac{\partial Q}{\partial x} - \frac{\partial P}{\partial y} \right) \cos \gamma \right] dS,$$

где $\cos \alpha$, $\cos \beta$, $\cos \gamma$ — направляющие косинусы нормали \vec{n} к поверхности σ . Формула Стокса утверждает, что циркуляция векторного поля \vec{a} по замкнутому контуру L равна потоку ротора векторного поля \vec{a} через любую кусочно-гладкую поверхность σ , натянутую на этот контур.

Примеры

Задача 9.32. Вычислить линейный интеграл векторного поля

$$\vec{a} = (x^2 + y^2 - 2px)\vec{i} + p(x + y)\vec{j}$$

вдоль дуги OA параболы $y^2 = px$, если точка $O(0, 0)$ и точка $A(p, p)$.

Решение. Линейный интеграл вычисляется по формуле

$$W = \int_{OA} \vec{a} \cdot d\vec{r} = \int_{OA} (x^2 + y^2 - 2px) dx + p(x + y) dy.$$

Используя уравнение параболы, получим $dx = \frac{2}{p} y dy$, после чего интеграл легко вычисляется:

$$\begin{aligned} W &= \int_0^p \left[\left(\frac{1}{p^2} y^4 + y^2 - 2y^2 \right) \frac{2}{p} y + p \left(\frac{1}{p} y^2 + y \right) \right] dy = \\ &= \left[\frac{2}{p^3} \frac{y^6}{6} - \frac{2}{p} \frac{y^4}{4} + \frac{y^3}{3} + \frac{p}{2} y^2 \right]_0^p = \frac{2}{3} p^3. \end{aligned}$$

В системе Maple эта задача решается очень просто

```
> a := VectorField( <x^2+y^2-2*p*x, p*(x+y)>, 'cartesian'[x,y] );
```

$$a := (x^2 + y^2 - 2px) \vec{e}_x + p(x + y) \vec{e}_y$$

```
> LineInt( a, Path( <t, (p*t)^(1/2)>, t=0..p ), inert ) =
```

```
LineInt( a, Path( <t, (p*t)^(1/2)>, t=0..p ) );
```

$$\int_0^p t^2 - p t + \frac{p^2 (t + \sqrt{p t})}{2 \sqrt{p t}} dt = \frac{1}{3} \operatorname{csgn}(p) p^3 + \frac{p^3}{3}$$

```
> assume(p>0):value(rhs(%));
```

$$\frac{2 p^3}{3}$$

Прокомментируем это решение. Как обычно, мы предполагаем уже подключенным пакет векторных вычислений `VectorCalculus`. В этом пакете имеется специальная команда для вычисления линейного интеграла, которая имеет формат

```
> LineInt(v, dom, inert);
```

Параметрами этой команды являются: `v` — определяет векторное поле, линейный интеграл которого вычисляется; `dom` — определяет путь интегрирования. Часто бывает полезно посмотреть на интеграл, который вычисляется. Для этого предусмотрена инертная форма. Если задан третий (необязательный) параметр `inert`, то интеграл не вычисляется, но выводится его начертание. Возможными путями интегрирования (параметр `dom`) могут быть следующие структуры:

- `Arc(obj, start, finish)` — дуга; первый параметр, `obj`, есть или `Circle`, или `Ellipse` структура; начальный и конечный углы дуги определяются параметрами `start` и `finish`;

- `Circle(cen, rad)` — окружность; `cen` — центр окружности (должен иметь тип `'Vector' (algebraic)`); `rad` — радиус (тип `algebraic`);
- `Circle3D(cen, rad, normal)` — окружность в пространстве; центр и радиус задаются так же как и для `Circle`; `normal` — нормаль к плоскости, в которой расположена окружность (тип `'Vector' (algebraic)`);
- `Ellipse(eq)` — эллипс, заданный уравнением `eq`;
- `Line(p1, p2)` — отрезок направленной прямой, соединяющей точки `p1, p2` (тип `'Vector' (algebraic)`, направление от `p1` к `p2`);
- `LineSegments(p1, p2, [p3, p4, ...])` — ломаная линия, состоящая из отрезков прямых `Line`;
- `Path(v, rng, [c])` — произвольный путь; `v` — вектор, задающий компоненты пути; `rng` должен иметь тип `{range, name=range}`; `c` — (необязательный) задает координатную систему, в которой определен вектор `v`.

Более подробную информацию, как обычно, можно получить по справке, например, набрав в строке знак вопроса и имя команды `?Path` и нажав клавишу ввода.

Вернемся к нашему примеру. Обратим внимание на то, что в результате вызова команды `LineInt` задача оказалась решенной не до конца. Это и естественно. Напомним, что Maple «по умолчанию» работает в самой широкой области комплексных чисел. Поэтому для корректного выполнения некоторых математических операций надо конкретизировать соответствующие переменные. В нашем случае необходимо вычислять корень квадратный, поэтому потребовалось определить знак переменной `p`.

Итак, окончательно, мы получили $\int_{OA} \vec{a} \cdot d\vec{r} = \frac{2}{3} p^3$.

Задача 9.33. Найти работу силового поля

$$\vec{F} = -y\vec{i} + x\vec{j} + k\vec{k}$$

по одному витку винтовой линии $L: x = a \cos t, y = a \sin t, z = bt$.

Решение. Работа силового поля выражается линейным интегралом

$$\int_L \vec{F} \cdot d\vec{r}.$$

В системе Maple решение задачи будет таким

```
> F:=VectorField(<-y,x,1>,'cartesian'[x,y,z]);
      F := -y  $\bar{\mathbf{e}}_x$  + x  $\bar{\mathbf{e}}_y$  +  $\bar{\mathbf{e}}_z$ 
> LineInt(F,Path(<a*cos(t),a*sin(t),b*t>,t=0..2*Pi),inert)=
      LineInt(F,Path(<a*cos(t),a*sin(t),b*t>,t=0..2*Pi));
```

$$\int_0^{2\pi} a^2 \sin(t)^2 + a^2 \cos(t)^2 + b dt = 2 a^2 \pi + 2 b \pi$$

> factor(rhs(%));

$$2 \pi (a^2 + b)$$

Итак, работа силового поля будет $A = 2\pi(a^2 + b)$.

Задача 9.34. Вычислить линейный интеграл векторного поля

$$\vec{a} = x\vec{i} + y\vec{j} + (x + y - 1)\vec{k}$$

по прямолинейному отрезку AB , где $A(1,1,1)$, $B(2,3,4)$.

Решение. Будем иметь

$$W = \int_{AB} \vec{a} \cdot d\vec{r} = \int_{AB} x dx + y dy + (x + y - 1) dz.$$

Выполним вычисления в Maple:

> SetCoordinates(cartesian[x,y,z]);

*cartesian*_{x,y,z}

> a:=VectorField(<x,y,x+y-1>);

$$a := x \vec{e}_x + y \vec{e}_y + (x + y - 1) \vec{e}_z$$

> LineInt(a, Line(<1,1,1>, <2,3,4>), inert)=

LineInt(a, Line(<1,1,1>, <2,3,4>));

$$\int_0^1 6 + 14 t dt = 13$$

Здесь первой командой мы сразу установили систему координат, в которой собираемся работать. Это бывает полезно, если мы не хотим указывать координаты всякий раз при обращении к другим командам.

Итак, искомый линейный интеграл равен 13.

Задача 9.35. Найти циркуляцию векторного поля

$$\vec{a} = -y^2\vec{i} + x^2\vec{j} + \vec{k}$$

вдоль замкнутой линии L , образованной пересечением сферы $x^2 + y^2 + z^2 = R^2$ с координатными плоскостями (рис. 9.6).




Рис. 9.6. К задаче 9.35

Решение. Данная линия состоит из трех дуг окружностей:

$$L_1: z = 0, x^2 + y^2 = R^2;$$

$$L_2: x = 0, y^2 + z^2 = R^2;$$

$$L_3: y = 0, x^2 + z^2 = R^2.$$

Следовательно,

$$C = \oint_L \vec{a} \cdot d\vec{r} = \oint_{L_1} \vec{a} \cdot d\vec{r} + \oint_{L_2} \vec{a} \cdot d\vec{r} + \oint_{L_3} \vec{a} \cdot d\vec{r}.$$

Все вычисления выполним в Maple.

```
> SetCoordinates( cartesian[x,y,z] );
                    cartesian
                    x,y,z
> a:=VectorField(<-y^2,x^2,1>);
                    a := -y^2 e_x + x^2 e_y + e_z
> LineInt( a, Path( <R*cos(t),R*sin(t),0>, t=0..Pi/2 ),
inert )+
LineInt( a, Path( <0,R*cos(t),R*sin(t)>, t=0..Pi/2 ),
inert )+
LineInt( a, Path( <R*cos(t),0,R*sin(t)>, t=Pi/2..Pi ),
inert )=
LineInt( a, Path( <R*cos(t),R*sin(t),0>, t=0..Pi/2 ) )+
LineInt( a, Path( <0,R*cos(t),R*sin(t)>, t=0..Pi/2 ) )+
LineInt( a, Path( <R*cos(t),0,R*sin(t)>, t=Pi/2..Pi ) );
```

$$\int_0^{\frac{\pi}{2}} R^3 \sin(t)^3 + R^3 \cos(t)^3 dt + \int_0^{\frac{\pi}{2}} R \cos(t) dt + \int_{\frac{\pi}{2}}^{\pi} R \cos(t) dt = \frac{4 R^3}{3}$$

Итак, искомая циркуляция

$$C = \oint_L \vec{a} \cdot d\vec{r} = \frac{4}{3} R^3 .$$

Задача 9.36. Вычислить циркуляцию вектора магнитной напряженности

$$\vec{H} = \frac{2J}{R^2} (-y\vec{i} + x\vec{j})$$

вдоль окружности $L: x^2 + y^2 = R^2, z = const$.

Решение. Имеем

$$C = \oint_L \vec{H} \cdot d\vec{r} = \frac{2J}{R^2} \oint_L (-ydx + xdy) .$$

> SetCoordinates(cartesian[x,y]);
cartesian_{x,y}

> H:=2*J/R^2*VectorField(<-y,x>);

$$H := -\frac{2 J y}{R^2} \vec{e}_x + \frac{2 J x}{R^2} \vec{e}_y$$

> LineInt(H,Path(<R*cos(t),R*sin(t)>,t=0..2*Pi), inert)=
LineInt(H,Path(<R*cos(t),R*sin(t)>,t=0..2*Pi));

$$\int_0^{2\pi} 2 J \sin(t)^2 + 2 J \cos(t)^2 dt = 4 \pi J$$

Итак, искомая циркуляция

$$C = \oint_L \vec{H} \cdot d\vec{r} = 4\pi J .$$

Задача 9.37. Найти циркуляцию векторного поля

$$\vec{a} = y\vec{i} - x\vec{j}$$

вдоль замкнутой линии $ABOA$ (рис. 9.7), где AB — дуга астроида, определяемой уравнением

$$x^{\frac{2}{3}} + y^{\frac{2}{3}} = R^{\frac{2}{3}} ,$$

или

$$x = R \cos^3 t, \quad y = R \sin^3 t.$$




Рис. 9.7. К задаче 9.37

Решение. Воспользуемся формулой Стокса. Тогда искомая циркуляция равна

$$C = \iint_{\sigma} \operatorname{rot} \vec{a} \cdot \vec{n} dS,$$

где σ — любая поверхность, натянутая на данный контур. Вычисляем $\operatorname{rot} \vec{a}$. Это удобно сделать в Maple:

$$\begin{aligned} > \operatorname{Curl}(\operatorname{VectorField}(\langle y, -x, 0 \rangle, \operatorname{cartesian}[x, y, z])); \\ & \quad (-2) \vec{e}_z \end{aligned}$$

Итак, $\operatorname{rot} \vec{a} = -2\vec{k}$. В качестве поверхности σ выберем плоскую область ABO . Тогда

$$C = \iint_{\sigma} (-2\vec{k}) \cdot \vec{k} dS = -2 \iint_{\sigma} dS = -2S,$$

где S — площадь области ABO . Вычисляем ее

$$S = \int_0^R y dx = 3R^2 \int_0^{\frac{\pi}{2}} \sin^4 t \cos^2 t dt = \frac{3}{32} \pi R^2.$$

Окончательно находим $C = -2S = -\frac{3}{16} \pi R^2$.

СОВЕТ

Интеграл тоже удобно вычислить в Maple:

$$> 3*R^2*\operatorname{Int}(\sin(t)^4*\cos(t)^2, t=0..Pi/2)=$$

$3R^2 \int_0^{\pi/2} \sin(t)^4 \cos(t)^2 dt$; $t=0..Pi/2$);

$$3 R^2 \int_0^{\frac{\pi}{2}} \sin(t)^4 \cos(t)^2 dt = \frac{3 R^2 \pi}{32}$$

Эту задачу в Maple можно решить и так

> SetCoordinates(cartesian[x,y]);

*cartesian*_{x,y}

> a:=VectorField(<y,-x>);

$$a := y \begin{matrix} \bar{e} \\ \mathbf{x} \end{matrix} - x \begin{matrix} \bar{e} \\ \mathbf{y} \end{matrix}$$

> LineInt(a, Path(<R*cos(t)^3,R*sin(t)^3>, t=0..Pi/2),

inert)+LineInt(a, Line(<0,R>, <0,0>), inert)+

LineInt(a, Line(<0,0>, <R,0>), inert)=

LineInt(a, Path(<R*cos(t)^3,R*sin(t)^3>, t=0..Pi/2))+

LineInt(a, Line(<0,R>, <0,0>))+

LineInt(a, Line(<0,0>, <R,0>));

$$\int_0^{\frac{\pi}{2}} -3 R^2 \sin(t)^4 \cos(t)^2 - 3 R^2 \cos(t)^4 \sin(t)^2 dt + \int_0^1 0 dt + \int_0^1 0 dt = -\frac{3 R^2 \pi}{16}$$

Второй способ вычисления линейного интеграла:

> LineInt(a, Path(<R*cos(t)^3,R*sin(t)^3>, t=0..Pi/2),

inert)+

LineInt(a, LineSegments(<0,R>, <0,0>, <R,0>), inert)=

LineInt(a, Path(<R*cos(t)^3,R*sin(t)^3>, t=0..Pi/2))+

LineInt(a, LineSegments(<0,R>, <0,0>, <R,0>));

$$\int_0^{\frac{\pi}{2}} -3 R^2 \sin(t)^4 \cos(t)^2 - 3 R^2 \cos(t)^4 \sin(t)^2 dt + 2 \int_0^1 0 dt = -\frac{3 R^2 \pi}{16}$$

Задача 9.38. Найти циркуляцию векторного поля

$$\vec{a} = xyzi\vec{i} + (x + y + z)\vec{j} - x^2y^2\vec{k}$$

вдоль контура квадрата (рис. 9.8), определяемого уравнениями: $x + y = a$, $x - y = a$, $x + y = -a$, $x - y = -a$, $z = 0$.




Рис. 9.8. К задаче 9.38

Решение. Выполним вычисление циркуляции заданного векторного поля двумя способами. Первый способ основан на формуле

$$C = \oint_{AB} \vec{a} \cdot d\vec{r} + \oint_{BC} \vec{a} \cdot d\vec{r} + \oint_{CD} \vec{a} \cdot d\vec{r} + \oint_{DA} \vec{a} \cdot d\vec{r}.$$

В системе Maple это выполняется так:

```
> SetCoordinates( cartesian[x,y,z] );
      cartesian
      x,y,z
```

```
> A:=VectorField(<x*y*z,x+y+z,-x^2*y^2>);
```

$$A := x y z \vec{e}_x + (x + y + z) \vec{e}_y - x^2 y^2 \vec{e}_z$$

```
> LineInt( A, LineSegments( <a,0,0>, <0,a,0>, <-a,0,0>,
  <0,-a,0>, <a,0,0> ), inert )=
```

```
LineInt( A, LineSegments( <a,0,0>, <0,a,0>, <-a,0,0>,
  <0,-a,0>, <a,0,0> ) );
```

$$\int_0^1 ((1-t)a + ta) a dt + \int_0^1 (-ta + (1-t)a) a dt + \int_0^1 -(1-t)a - ta) a dt + \int_0^1 (ta - (1-t)a) a dt = 2a^2$$

Таким образом, $C = 2a^2$.

Второй способ основан на применении формулы Стокса, тогда

$$C = \iint_{\sigma} \text{rot } \vec{a} \cdot \vec{n} dS,$$

где σ — любая поверхность, натянутая на данный контур. Вычисляем $\operatorname{rot} \vec{a}$. Это удобно сделать в Maple:

> Curl(A);

$$(-2x^2y - 1)\vec{e}_x + (xy + 2xy^2)\vec{e}_y + (1 - xz)\vec{e}_z$$

Итак, $\operatorname{rot} \vec{a} = -(2x^2y^2 + 1)\vec{i} + (2xy^2 + xy)\vec{j} + (1 - xz)\vec{k}$. В качестве поверхности σ выбираем квадрат $ABCD$. Вектор нормали, фигурирующий в формуле Стокса $\vec{n} = \vec{k}$. Следовательно, $\vec{k} \cdot \operatorname{rot} \vec{a} = 1 - xz|_{z=0} = 1$ и

$$C = \iint_{\sigma} dS = S_{ABCD} = 2a^2.$$

Потенциальное и соленоидальное векторные поля

Простейшими векторными полями являются такие, для которых либо $\operatorname{div} \vec{a}(M) = 0$, либо $\operatorname{rot} \vec{a}(M) = 0$. Векторное поле $\vec{a}(M)$, в котором $\operatorname{rot} \vec{a}(M) = 0$, называется *безвихревым* или *потенциальным*. Векторное поле $\vec{a}(M)$, в котором $\operatorname{div} \vec{a}(M) = 0$, называется *трубчатым* или *соленоидальным*.

Можно также сказать, что векторное поле $\vec{a}(M)$ называется потенциальным в области V , если существует скалярная функция $u(M)$, заданная в области V , такая, что для всех точек области $\vec{a}(M) = \operatorname{grad} u(M)$. Функция $u(M)$ называется *потенциалом* поля $\vec{a}(M)$.

Аналогично, можно сказать, что векторное поле $\vec{a}(M)$ называется соленоидальным в области V , если существует такое векторное поле $\vec{A}(M)$, что всюду в области $\vec{a}(M) = \operatorname{rot} \vec{A}(M)$. Поле $\vec{A}(M)$ называется *вектором-потенциалом* (или векторным потенциалом) поля $\vec{a}(M)$.

Условие $\operatorname{rot} \vec{a}(M) = 0$ является необходимым условием потенциальности поля $\vec{a}(M)$. Достаточным условием потенциальности поля $\vec{a}(M)$ служит следующее: если $\operatorname{rot} \vec{a}(M) = 0$ и область V односвязная, то поле $\vec{a}(M)$ — потенциальное. Для потенциального поля имеет место равенство

$$\int_L \vec{a} \cdot d\vec{r} = u(M_1) - u(M_0),$$

показывающее, что интеграл в левой части не зависит от выбора пути L и определяется только начальной M_0 и конечной M_1 точками пути. Кроме того, из последнего равенства следует, что потенциал $u(M)$ может быть найден по формуле

$$u(M) = u(M_0) + \int_{M_0}^M \vec{a} \cdot d\vec{r},$$

где M_0 — фиксированная точка и $u(M_0)$ — любая постоянная.

Если $\vec{a} = P\vec{i} + Q\vec{j} + R\vec{k}$, а в качестве пути, соединяющего точки M_0 и M_1 , берется ломаная из трех звеньев, параллельных координатным осям, то

$$u(x, y, z) = \int_{x_0}^x P(x, y_0, z_0) dx + \int_{y_0}^y Q(x, y, z_0) dy + \int_{z_0}^z R(x, y, z) dz.$$

Условие $\operatorname{div} \vec{a}(M) = 0$ является необходимым условием соленоидальности векторного поля $\vec{a}(M)$. Достаточным условием служит следующее: если $\operatorname{div} \vec{a}(M) = 0$ и область V односвязная, то поле $\vec{a}(M)$ — соленоидальное.

Примеры

Задача 9.39. Доказать, что потенциал векторного поля \vec{a} определяется с точностью до аддитивной постоянной, а векторный потенциал определяется с точностью до слагаемого вида $\operatorname{grad} u$.

Решение. Пусть \vec{a} — потенциальное поле и u_1, u_2 — два различных потенциала, то есть $\vec{a} = \operatorname{grad} u_1$ и $\vec{a} = \operatorname{grad} u_2$. Отсюда следует $\operatorname{grad}(u_1 - u_2) = 0$, то есть $u_1 - u_2 = \operatorname{const}$, что и требовалось доказать.

Пусть \vec{a} — соленоидальное поле и \vec{A}_1, \vec{A}_2 — два различных векторных потенциала, то есть $\vec{a} = \operatorname{rot} \vec{A}_1$ и $\vec{a} = \operatorname{rot} \vec{A}_2$. Отсюда следует $\operatorname{rot}(\vec{A}_1 - \vec{A}_2) = 0$. А отсюда (если область односвязная) $\vec{A}_1 - \vec{A}_2 = \operatorname{grad} u$, что и требовалось доказать.

Задача 9.40. Проверить, что поле $\vec{a} = (y+z)\vec{i} + (z+x)\vec{j} + (x+y)\vec{k}$ является потенциальным и найти его потенциал.

Решение. Поле \vec{a} определено во всем пространстве, то есть в односвязной области, поэтому достаточно проверить, что $\operatorname{rot} \vec{a}(M) = 0$. Это легко сделать в Maple:

```
> a:=VectorField(<y+z, z+x, x+y>, 'cartesian'[x, y, z]);
      a := (y + z)  $\vec{e}_x$  + (z + x)  $\vec{e}_y$  + (x + y)  $\vec{e}_z$ 
> Curl(a);
      0  $\vec{e}_x$ 
```

Для нахождения потенциала воспользуемся формулой

$$u(x, y, z) = \int_{x_0}^x P(x, y_0, z_0) dx + \int_{y_0}^y Q(x, y, z_0) dy + \int_{z_0}^z R(x, y, z) dz,$$

беря в качестве точки M_0 начало координат, то есть $x_0 = 0$, $y_0 = 0$, $z_0 = 0$. Будем иметь

$$\begin{aligned} u(x, y, z) &= \int_0^x P(x, 0, 0) dx + \int_0^y Q(x, y, 0) dy + \int_0^z R(x, y, z) dz = \\ &= \int_0^x 0 dx + \int_0^y x dy + \int_0^z (x + y) dz = xy + yz + zx. \end{aligned}$$

В Maple эта задача решается так:

```
> LineInt( a, Line( <0,0,0>, <x,y,z> ), inert )=
  LineInt( a, Line( <0,0,0>, <x,y,z> ) );
  ∫₀¹ (ty + tz)x + (tz + tx)y + (tx + ty)z dt = (y+z)x/2 + (z+x)y/2 + (x+y)z/2
> simplify(rhs(%));
  x y + x z + y z
```

то есть с помощью формулы

$$u(M) = \int_{M_0}^M \vec{a} \cdot d\vec{r},$$

где M_0 — фиксированная точка.

Итак, искомым потенциал $u = xy + yz + zx$.

Дифференциальные операции второго порядка

К дифференциальным операциям первого порядка относятся скалярное поле $\operatorname{div} \vec{a}$ и векторные поля $\operatorname{grad} u$ и $\operatorname{rot} \vec{a}$.

Возьмем от скалярного поля $\operatorname{div} \vec{a}$ градиент, а от векторных полей $\operatorname{grad} u$ и $\operatorname{rot} \vec{a}$ — дивергенцию и ротор. Тогда получим пять дифференциальных операций второго порядка:

1. $\operatorname{grad}(\operatorname{div} \vec{a})$,
2. $\operatorname{div}(\operatorname{grad} u)$,
3. $\operatorname{rot}(\operatorname{grad} u)$,

$$4. \operatorname{div}(\operatorname{rot} \vec{a}),$$

$$5. \operatorname{rot}(\operatorname{rot} \vec{a}).$$

В символической форме эти операции запишутся так:

$$1. \operatorname{grad}(\operatorname{div} \vec{a}) = \nabla(\nabla \cdot \vec{a}),$$

$$2. \operatorname{div}(\operatorname{grad} u) = \nabla \cdot (\nabla u),$$

$$3. \operatorname{rot}(\operatorname{grad} u) = \nabla \times (\nabla u),$$

$$4. \operatorname{div}(\operatorname{rot} \vec{a}) = \nabla \cdot (\nabla \times \vec{a}),$$

$$5. \operatorname{rot}(\operatorname{rot} \vec{a}) = \nabla \times (\nabla \times \vec{a}).$$

Отметим, что третья и четвертая операции дают тождественный нуль (Проверьте!). Вторая из указанных операций дает оператор Лапласа

$$\operatorname{div}(\operatorname{grad} u) = \nabla \cdot (\nabla u) = \nabla^2 u = \Delta u.$$

Между первой и пятой операциями существует связь

$$\operatorname{rot}(\operatorname{rot} \vec{a}) = \operatorname{grad}(\operatorname{div} \vec{a}) - \Delta \vec{a}.$$

Примеры

Задача 9.41. Доказать, что $\operatorname{rot}(\operatorname{grad} u) = 0$.

Решение. В системе Maple:

```
> SetCoordinates( cartesian[x,y,z] );
```

cartesian
x,y,z

```
> Gradient(u(x,y,z));
```

$$\left(\frac{\partial}{\partial x} u(x,y,z)\right) \vec{e}_x + \left(\frac{\partial}{\partial y} u(x,y,z)\right) \vec{e}_y + \left(\frac{\partial}{\partial z} u(x,y,z)\right) \vec{e}_z$$

```
> Curl(%);
```

$$0 \vec{e}_x$$

Формально равенство $\operatorname{rot}(\operatorname{grad} u) = 0$ можно получить символическим методом:

$$\operatorname{rot}(\operatorname{grad} u) = \nabla \times (\nabla u) = \nabla \times \nabla u = 0,$$

так как векторное произведение двух равных векторов тождественно равно нулю.

Задача 9.42. Вычислить $\operatorname{div}(\operatorname{grad} u)$, где $u = r^2$.

Решение. В системе Maple:

```
> SetCoordinates( cartesian[x,y,z] );  
                    cartesian  
                    x,y,z  
  
> Gradient(u(x,y,z));  
      (∂/∂x u(x,y,z)) e_x + (∂/∂y u(x,y,z)) e_y + (∂/∂z u(x,y,z)) e_z  
  
> Divergence(%);  
      (∂²/∂x² u(x,y,z)) + (∂²/∂y² u(x,y,z)) + (∂²/∂z² u(x,y,z))  
  
> u := (x,y,z) -> x^2+y^2+z^2;  
      u := (x,y,z) -> VectorCalculus :-`+`(VectorCalculus :-`+`(x^2,y^2),z^2)  
  
> %%;
```

6

Итак, $\operatorname{div}(\operatorname{grad} r^2) = 6$. Мы здесь попутно еще раз доказали, что $\operatorname{div}(\operatorname{grad} u) = \Delta u$.

Упражнения

В задачах 1 – 5 установить область определения и найти поверхности (линии) уровня скалярного поля:

1. $u = \sqrt{7 - x^2 - y^2}$.

Ответ. Область определения — круг $x^2 + y^2 \leq 7$; линии уровня — семейство концентрических окружностей $x^2 + y^2 = 7 - C^2$, $|C| \leq \sqrt{7}$.

2. $u = \frac{q}{r}$ (потенциал электростатического поля).

Ответ. Поле определено во всем пространстве, за исключением точки $r = 0$; поверхности уровня — сферы $r = C$ с центром в точке, где находится заряд q .

3. $u = \arcsin\left(\frac{x}{\sqrt{y^2 + z^2}}\right)$.

Ответ. Поле определено в области $y^2 + z^2 - x^2 \geq 0$; поверхности уровня — круговые конусы $a^2(y^2 + z^2) - x^2 = 0$ ($|a| \leq 1$).

4. $u = \sin(x^2 - y^2)$.

Ответ. Линия уровня $u = C$ представляет собой семейство гипербол $x^2 - y^2 = (-1)^n \arcsin C + n\pi$, где n — любое целое число.

5. $u = \frac{x^2 + y^2}{z}$.

Ответ. Поле определено во всем пространстве, за исключением плоскости $z = 0$. Поверхности уровня — параболоиды вращения $x^2 + y^2 = Cz$ ($-\infty < C < +\infty$).

6. Найти производную плоского скалярного поля $u = x^2 + y^2 - 3x + 2y$ в начале координат по направлению, идущему из начала координат в точку $P(3, 4)$.

Ответ. $\frac{\partial u}{\partial l} = -\frac{1}{5}$.

7. Найти производную скалярного поля $u = xy^2 + z^3 - xyz$ в точке $M(1, 1, 2)$ в направлении, образующем с осями координат углы 60° , 45° и 60° соответственно.

Ответ. $\frac{\partial u}{\partial l} = 5$.

8. Найти производную скалярного поля $u = \ln(x^2 + y^2)$ в точке $P_0(x_0, y_0)$ в направлении, перпендикулярном к линии уровня функции u , проходящей через эту точку.

Ответ. $\frac{\partial u}{\partial l} = \frac{2}{r}$, где $r = \sqrt{x_0^2 + y_0^2}$ — расстояние точки P_0 от начала координат.

9. Найти производные поля $u = zy^2 + z^2 - 2xyz$ в точке $P_0(3, 1, 1)$ в направлении координатных осей.

Ответ. По направления оси Ox производная равна -2 ; по направления оси Oy производная равна -4 ; по направления оси Oz производная равна -3 .

10. Найти производную скалярного поля $u = \ln(x + y)$ в точке $P_0(1, 2)$ по направлению параболы $y^2 = 4x$, проходящей через эту точку.

Ответ. $\frac{\partial u}{\partial l} = \frac{\sqrt{2}}{3}$.

11. Исследовать характер изменения поля $u = 5yzx^2 + 5xyz^2 - 7xy^2z$ в направлении вектора $\vec{a} = 8\vec{i} - 4\vec{j} + 8\vec{k}$ в точке $P_0(1, 1, 1)$ и найти величину скорости этого изменения.

Ответ. Поле возрастает со скоростью $v = 12$.

12. Найти производную скалярного поля $u(P) = 2xy + 2z$ по направлению винтовой линии $x = 3 \cos t$, $y = 3 \sin t$, $z = 4t$ в точке N , где $t = \frac{3\pi}{2}$.

Ответ. $\frac{\partial u}{\partial l} = -2$.

13. Найти производную скалярного поля $u(P) = x^2 + yz$ по направлению винтовой линии $x = 4 \cos t$, $y = 4 \sin t$, $z = 3t$ в точке N , где $t = \pi$.

Ответ. $\frac{\partial u}{\partial l} = \frac{2}{5}(8 + 9\pi)$.

14. Найти производную плоского поля $u(P) = x^3 + xy + 3y^4$ в точке $M_0(-2, 1, 0)$ по направлению, лежащему в плоскости xOy и наклоненному под углом $\frac{\pi}{6}$ к оси Ox .

Ответ. $\frac{\partial u}{\partial l} = \frac{13}{2}\sqrt{3} + 5$.

15. Найти производную скалярного поля $u(P) = xz^2 + 2yz$ в точке $M_0(1, 0, 2)$ вдоль окружности $\vec{r}(t) = (\cos t + 1)\vec{i} + (\sin t - 1)\vec{j} + 2\vec{k}$.

Ответ. $\frac{\partial u}{\partial l} = -4$.

16. Найти градиент скалярного поля $u = x^2 + 2y^2 + 3z^2 - xy - 4x + 2y - 4z$ в начале координат.

Ответ. $\text{grad } u = -4\vec{i} + 2\vec{j} - 4\vec{k}$.

17. Найти градиент скалярного поля $u = 3x^2y - 3y^3 + y^4$ в точке $M(1, 2)$.

Ответ. $\text{grad } u = 12\vec{i} - \vec{j}$.

18. Найти градиент скалярного поля $u = \sqrt{x^2 + y^2 + z^2} = r$.

Ответ. $\text{grad } u = \frac{\vec{r}}{r}$.

19. Найти градиент скалярного поля $u = \text{arctg}\left(\frac{y}{x}\right)$.

Ответ. $\text{grad } u = -\frac{y}{x^2 + y^2}\vec{i} + \frac{x}{x^2 + y^2}\vec{j}$.

20. Найти наибольшую скорость v возрастания поля $u = \ln(x^2 + 4y^2)$ в точке $M(6, 4)$.

Ответ. $v = \frac{2\sqrt{73}}{5}$.

21. Найти наибольшую скорость v возрастания поля $u = x^2 - y^2 + x + y$ в точке $M(1, 1)$.

Ответ. $v = \sqrt{10}$.

22. Найти $\text{grad}\left(\frac{1}{r^3}\right)$.

Ответ. $\text{grad } u = -3\frac{\vec{r}}{r^5}$.

23. Найти $\text{grad}(r^n)$, $n > 0$.

Ответ. $\text{grad } u = nr^{n-2}\vec{r}$.

24. Определить единичный вектор нормали к сфере $x^2 + y^2 + z^2 = R^2$.

Ответ. $\vec{n} = \frac{x\vec{i} + y\vec{j} + z\vec{k}}{R}$.

25. Доказать равенство $\text{grad}\left(\frac{\varphi}{\psi}\right) = \frac{1}{\psi^2}[\psi \text{grad}(\varphi) - \varphi \text{grad}(\psi)]$, где $\varphi(x, y, z)$ и $\psi(x, y, z)$ — скалярные функции.

26. Вычислить поток вектора $\vec{a} = (x - y)\vec{i} + (3y - z)\vec{j} + (2z - x)\vec{k}$ через треугольник, получаемый при пересечении плоскости $6x + 2y + 3z - 6 = 0$ с плоскостями координат.

Ответ. $\Pi = \frac{11}{6}$.

27. Вычислить поток вектора $\vec{a} = (x - 2y)\vec{i} + z\vec{j} + (3y + z)\vec{k}$ через часть сферической поверхности единичного радиуса с центром в начале координат, расположенную в первом октанте.

Ответ. $\Pi = \frac{\pi + 2}{3}$.

28. Вычислить поток вектора $\vec{a} = xy\vec{i} + yz\vec{j} + xz\vec{k}$ через часть сферы $x^2 + y^2 + z^2 = 1$, расположенную в первом октанте.

Ответ. $\Pi = \frac{3}{16}$.

29. Вычислить поток вектора $\vec{a} = xy\vec{i} + y\vec{j} - z\vec{k}$ через полную поверхность сферы $x^2 + y^2 + z^2 = 4$.

Ответ. $\Pi = 0$.

30. Вычислить с помощью формулы Остроградского поток вектора $\vec{a} = 3x\vec{i} - y\vec{j} + z\vec{k}$ через полную поверхность конуса $9 - z = x^2 + y^2$, $0 \leq z < 9$.

Ответ. $\Pi = 81\pi$.

31. Вычислить с помощью формулы Остроградского поток вектора $\vec{a} = x^3\vec{i} + y^3\vec{j} + z^3\vec{k}$ через полную поверхность сферы $x^2 + y^2 + z^2 = R^2$.

Ответ. $\Pi = \frac{12\pi R^5}{5}$.

32. Найти поток радиус-вектора $\vec{r} = x\vec{i} + y\vec{j} + z\vec{k}$ через замкнутую поверхность S , ограничивающую объем V .

Ответ. $\Pi = 3V$.

33. Найти с помощью формулы Остроградского поток вектора $\vec{a} = y^2\vec{i} + z^2\vec{j} + z^2\vec{k}$ через полную поверхность эллипсоида $\frac{x^2}{a^2} + \frac{y^2}{b^2} + \frac{z^2}{c^2} = 1$.

Ответ. $\Pi = 0$.

34. Вычислить поток вектора $\vec{a} = 3x\vec{i} - 2z\vec{j} + y\vec{k}$ через полную поверхность пирамиды $x + y + z = 2$, $x \geq 0$, $y \geq 0$, $z \geq 0$.

Ответ. $\Pi = 4$.

35. Вычислить поток вектора $\vec{a} = 3x\vec{i} + 3y\vec{j} - z\vec{k}$ через поверхность $x^2 + y^2 = a^2$, $0 \leq z \leq H$.

Ответ. $\Pi = 5\pi a^2 H$.

36. Вычислить дивергенцию векторного поля $\vec{a} = \alpha\vec{r}$, где $\alpha = const$.

Ответ. $\operatorname{div} \vec{a} = 3\alpha$.

37. Вычислить дивергенцию векторного поля $\vec{a} = (4x^2z + 3)\vec{i} + (2xy - 3)\vec{j} + (6xyz - 9)\vec{k}$.

Ответ. $\operatorname{div} \vec{a} = 2x(4z + 3y + 1)$.

38. Вычислить дивергенцию векторного поля $\vec{a} = r\vec{c}$, где \vec{c} — постоянный вектор.

Ответ. $\operatorname{div} \vec{a} = \frac{\vec{r} \cdot \vec{c}}{r}$.

39. Вычислить дивергенцию векторного поля $\vec{a} = r^4\vec{r}$.

Ответ. $\operatorname{div} \vec{a} = 7r^4$.

40. Вычислить дивергенцию векторного поля $\vec{a} = \vec{b}(\vec{r} \cdot \vec{c})$, где \vec{b} и \vec{c} — постоянные векторы.

Ответ. $\operatorname{div} \vec{a} = \vec{b} \cdot \vec{c}$.

41. Вычислить дивергенцию векторного поля $\vec{a} = \vec{r} \times (\vec{c} \times \vec{r})$, где \vec{c} — постоянный вектор.

Ответ. $\operatorname{div} \vec{a} = -2\vec{r} \cdot \vec{c}$. *Указание.* Воспользоваться формулой двойного векторного произведения $\vec{A} \times (\vec{B} \times \vec{C}) = \vec{B}(\vec{A} \cdot \vec{C}) - \vec{C}(\vec{A} \cdot \vec{B})$.

42. Найти дивергенцию градиента функции $u = e^{x+y+z}$.

Ответ. $\operatorname{div}(\operatorname{grad} u) = 3u$.

43. Найти $\operatorname{div}(\vec{u} \times \vec{v})$, где $u = x\vec{i} + y\vec{j} + z\vec{k}$, $v = y\vec{i} + z\vec{j} + x\vec{k}$.

Ответ. $\operatorname{div}(\vec{u} \times \vec{v}) = x + y + z$.

44. Вычислить ротор векторного поля $\vec{a} = xyz\vec{r}$, где $\vec{r} = x\vec{i} + y\vec{j} + z\vec{k}$ — радиус-вектор точки.

Ответ. $\operatorname{rot} \vec{a} = x(z^2 - y^2)\vec{i} + y(x^2 - z^2)\vec{j} + z(y^2 - x^2)\vec{k}$.

45. Вычислить ротор векторного поля $\vec{a} = \frac{\vec{r}}{r}$, где $\vec{r} = x\vec{i} + y\vec{j} + z\vec{k}$, $r = |\vec{r}|$.

Ответ. $\operatorname{rot} \vec{a} = 0$.

46. Вычислить линейный интеграл векторного поля $\vec{a} = (x^2 + y^2 - 2Rx)\vec{i} + R(x + y)\vec{j}$ вдоль дуги OA окружности $(x - R)^2 + y^2 = R^2$, где $O(0,0)$ и $A(R, R)$.

Ответ. $W = \frac{6 - \pi}{4} R^3$.

47. Вычислить работу силового поля $\vec{f} = -(x\vec{i} + y\vec{j})$ вдоль дуги AB , $A(a, 0)$,

$B(0, b)$ эллипса $\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$, лежащей в первом квадранте.

Ответ. $W = \frac{a^2 - b^2}{2}$.

48. Найти циркуляцию поля $\vec{a} = y\vec{i}$ по контуру окружности $x = b \cos t$,
 $y = b + b \sin t$, расположенной в плоскости xOy .

Ответ. $C = -\pi b^2$.

49. Вычислить двумя способами (непосредственно и с помощью формулы Стокса)
циркуляцию поля $\vec{a} = x^2 y^3 \vec{i} + \vec{j} + z \vec{k}$ вдоль окружности $x^2 + y^2 = R^2$, $z = 0$.

Ответ. $C = -\frac{1}{8} \pi R^6$.

50. Вычислить двумя способами (непосредственно и с помощью формулы Стокса)
циркуляцию поля $\vec{a} = \frac{\vec{r}}{r}$ вдоль контура $(x - 2)^2 + (y - 2)^2 + (z - 2)^2 = 1$, $x = \frac{3}{2}$.

Ответ. $C = 0$.

51. Проверить, что поле $\vec{a} = yz\vec{i} + zx\vec{j} + xy\vec{k}$ является потенциальным и найти его потенциал.

Ответ. $u = xyz$.

52. Доказать, что $\operatorname{div}(\operatorname{rot} \vec{a}) = 0$.

53. Доказать тождество $\operatorname{rot}(\operatorname{rot} \vec{a}) = \operatorname{grad}(\operatorname{div} \vec{a}) - \Delta \vec{a}$.

Литература

1. Будак Б.М., Самарский А.А., Тихонов А.Н. Сборник задач по математической физике. – М.: ГИТТЛ, 1956. – 684 с.
2. Владимиров В.С., Михайлов В.П., Вашарин А.А., Каримова Х.Х., Сидоров Ю.В., Шабунин М.И. Сборник задач по уравнениям математической физики. – Изд. 2-е. – М.: Наука, 1982. – 256 с.
3. Голоскоков Д.П. Уравнения математической физики. Решение задач в системе Maple. Учебник для вузов. – СПб.: Питер, 2004. – 539 с.
4. Кручкович Г.И., Мордасова Г.М., и др. Сборник задач и упражнений по специальным главам высшей математики. – М.: Высшая школа, 1970. – 512 с.
5. Лебедев Н.Н., Скальская И.П., Уфлянд Я.С. Сборник задач по математической физике. – М.: ГИТТЛ, 1959. – 420 с.
6. Михлин С.Г. Вариационные методы в математической физике. – М.: ГИТТЛ, 1957 – 476 с.
7. Очан Ю.С. Методы математической физики. – М.: Высшая школа, 1965. – 384 с.
8. Зенкевич О., Морган К. Конечные элементы и аппроксимация. – М.: Мир, 1986. – 318 с.
9. Семянистый В.И. Задачник-практикум по курсу «Методы математической физики». – М., Просвещение, 1965 – 165 с.

Учебное издание

Голоскоков Дмитрий Петрович

**ПРАКТИЧЕСКИЙ КУРС
МАТЕМАТИЧЕСКОЙ ФИЗИКИ
В СИСТЕМЕ MAPLE**


Учебное пособие

Редактор *Э. А. Горелик*

Подписано в печать 25.01.10	Сдано в производство 25.01.10	
Формат 60x84 1/16	Усл.-печ. л. 74,24.	Уч.-изд. л. 64.
Тираж 5000 экз.	Заказ № 3207	

**Издательство «ПаркКом»
Санкт-Петербург, ул. Наличная, 17**


**Отпечатано в типографии ООО «ПаркКом»
Санкт-Петербург, ул. Наличная, 17**



**Голоскоков Дмитрий Петрович —
доктор технических наук, профессор,
заведующий кафедрой прикладной
математики
Санкт-Петербургского государственного
университета водных коммуникаций**

«Книга представляет собой редкий в наше время авторский курс изложения математики с использованием пакета Maple. Курс отличают простые и понятные определения, наглядные примеры, продуманные задачи, четкие рекомендации для пользователей. Читая книгу, рецензент нашел для себя много нового и полезного. В настоящее время в русскоязычной литературе по теме Maple существует не более 5-10 книг такого уровня, поэтому книга Голоскокова Д.П. весьма актуальна, востребована и, можно сказать, долгожданна».

**Кирсанов М.Н.
доктор физико-математических наук,
профессор МЭИ (ТУ)**



ISBN 978-5-98175-034-0



9 785981 750342